Curiosités et traduction pour
`Xcas`

Renée De Graeve

17 novembre 2018

Chapitre 1

Pour s'amuser en arithmétique

1.1 Calcul pour $p = 1..7$ des sommes $\sum_{k=1}^{n} k^p$

1.1.1 Calculer $S_1 = \sum_{k=1}^{n} k$

On pose $S_1 = \sum_{k=1}^{n} k$.
S_1 a n termes.
On a :
$S_1 = 1 + \quad 2 + \quad 3 + ... + (n-2) + (n-1) + n$ et
$S_1 = n + (n-1) + (n-2) + ... + \quad 3 + \quad 2 + 1$
Donc en additionnant terme à terme on a :
$2S_1 = 1+n) + (2+n-1) + (3+n-2) + ... + ((n-2+3) + (n-1+2) + n+1)$
$2S_1$ a n termes égaux à $n+1$ donc :
$2S_1 = n(n+1)$ donc

$$S_1 = \frac{n(n+1)}{2}$$

Avec `Xcas`.
On tape :
`factor(sum(k,k=1..n))`
On obtient :
`n*(1+n)/2`

1.1.2 Calculer $S_2 = \sum_{k=1}^{n} k^2$

On sait que :
$k^3 = ((k-1)+1)^3 = (k-1)^3 + 3(k-1)^2 + 3(k-1) + 1$
Donc on a :
$(n+1)^3 = n^3 + 3n^2 + 3n + 1$
$n^3 \quad = (n-1)^3 + 3(n-1)^2 + 3(n-1) + 1$
$(n-1)^3 = (n-2)^3 + 3(n-2)^2 + 3(n-2) + 1$
....
$3^3 \quad = 2^3 + 3*2^2 + 3*2 + 1$
$2^3 \quad = 1^3 + 3*1^2 + 3*1 + 1$
$1^3 \quad = 1$
En additionnant ces égalites on obtient :
$(n+1)^3 = 3\sum_{k=1}^{n} k^2 + 3\sum_{k=1}^{n} k + (n+1)*1$

Donc :
$3\sum_{k=1}^{n} k^2 = (n+1)^3 - 3\frac{n(n+1)}{2} - (n+1)$
$3\sum_{k=1}^{n} k^2 = (n+1)(n^2+2n+1-\frac{3(n+1)}{2}-1)$
$\sum_{k=1}^{n} k^2 = (n+1)\frac{2n^2+n}{6}$ donc

$$S_2 = \frac{n(n+1)(2n+1)}{6}$$

Avec `Xcas`.
On tape :
`factor(sum(k^2,k=1..n))`
On obtient :
`n*(1+n)*(1+2*n)/6`

1.1.3 Calculer $\sum_{k=1}^{n} k^p$ pour $p = 3..7$

Pour calculer les sommes $S_p = \sum_{k=1}^{n} k^p$, on utilise la même technique.
Pour calculer $S_3 = \sum_{k=1}^{n} k^3$ on utilise :
$k^4 = ((k-1)+1)^4 = (k-1)^4 + 4(k-1)^3 + 6(k-1)^2 + 4(k-1) + 1$
Pour calculer $S_4 = \sum_{k=1}^{n} k^4$ on utilise :
$k^5 = ((k-1)+1)^5 = (k-1)^5+5(k-1)^3+10(k-1)^3+10(k-1)^2+5(k-1)+1$
etc..
Avec `Xcas`.
On tape :
`factor(sum(k^3,k=1..n))`
On obtient :
`n^2*(1+n)^2/4`
Donc

$$S_3 = S_1^2 = \frac{n^2(n+1)^2}{4}$$

On tape :
`factor(sum(k^4,k=1..n))`
On obtient :
`n*(1+n)*(1+2*n)*(-1+3*n+3*n^2)/60`
On tape :
`factor(sum(k^5,k=1..n))`
On obtient :
`n^2*(1+n)^2*(-1+2*n+2*n^2)/12`
On tape :
`factor(sum(k^6,k=1..n))`
On obtient :
`n*(1+n)*(1+2*n)*(1-3*n+6*n^3+3*n^4)/42`
On tape :
`factor(sum(k^7,k=1..n))`
On obtient :
`n^2*(1+n)^2*(2-4*n-n^2+6*n^3+3*n^4)/24`

1.1.4 Calculer $\sum_{k=1}^{2016} k^5 + k^7$

Avec `Xcas`.
On tape :
`factor(sum(k^7,k=1..n)-sum(k^5,k=1..n))`
On obtient :
`n^4*(1+n)^4/8`
On remarque que :
$S_7 - S_5 = 2S_1^4$
Donc pour $n = 2016$ on a :
$S_5+S_7 = 2016^4*2017^4/8 = 34173992277321737788588032 \simeq 3.4173992e+25$

1.2 Calcul de $9999998^2 - 9999999 * 9999997$

1.2.1 Le calcul

Calculer $9999998^2-9999999*9999997$ et plus généralement calculer $(9...98)^2-(9...99) * (9...97)$ pour des nombres ayant p chiffres.
On pose $x = 10^7$ (et plus généralement $x = 10^p$).
On doit donc calculer : $(x-2)^2-(x-1)*(x-3) = (x^2-4x+4-(x^2-4x+3) = 1$
Donc $9999998^2 - 9999999 * 9999997 = 1$ On calcule avec `Xcas`, on tape :
`9999998^2-9999999*9999997`
On obtient :
`1`
On tape :
`normal((x-2)^2-(x-1)*(x-3))`
On obtient :
`1`

1.2.2 Trouver un exemple du même type

On a par exemple :
$(x - 3)^2 - (x - 2) * (x - 4) = 1$
$(x - 4)^2 - (x - 3) * (x - 5) = 1$
$(x - 10)^2 - (x - 8) * (x - 12) = 1$
$(x-a)^2-(x-a+1)*(x-a-1) = x^2-2ax+a^2-(x^2-2ax+(a-1)(a+1)) = 1$
Donc on peut calculer par exemple avec $x = 10^7$ et $a = 12$:
$9999988^2 - 9999989 * 9999987 = 1$
ou encore avec $x = 10^7$ et $a = 102$
$9999898^2 - 9999899 * 9999897 = 1$ On peut aussi utiliser légalité calculer :
$(x-a)^2-(x-a+b)*(x-a-b) = x^2-2ax+a^2-(x^2-2ax+(a-b)(a+b)) = b^2$
puis par exemple, calculer avec $x = 10^7$ et $(a, b) = (5, 4)$:
$9999995^2 - 9999999 * 9999991 = 16$

1.3 Calcul de $55555556^2 - 44444445^2$

1.3.1 Le calcul

Calculer :
$55555556^2 - 44444445^2$ et plus généralement calculer $(5...56)^2 - (4...45)^2$ pour des nombres ayant p chiffres.

On calcule avec `Xcas` :
$6^2 - 5^2 = 11$ (l'écriture décimale comporte 2 fois le chiffre un),
$56^2 - 45^2 = 1111$ (l'écriture décimale comporte 4 fois le chiffre un),
....
$55555556^2 - 44444445^2 = 1111111111111111$ (l'écriture décimale comporte 16 fois le chiffre un).
Il semble donc que l'on a à démontrer que :
$(5...56)^2 - (4...45)^2$ pour des nombres ayant p chiffres vaut un nombre dont l'écriture décimale comporte $2p$ fois le chiffre un.
Pour le démontrer, on va simplement d'utiliser l'identité remarquable :
$a^2 - b^2 = (a - b)(a + b)$
On a ($p = 8$) :
$55555556^2 - 44444445^2 = (55555556 + 44444445)(55555556 - 44444445) = 100000001 * 11111111 = 1111111111111111$.
Plus généralement, si on pose x_p le nombre qui s'écrit avec p fois le chiffre 1, on doit calculer :
$(5 * x_p + 1)^2 - (4 * x_p + 1)^2 = (9 * x_p + 2)(x_p)$.
On a $x_p = 11...1 = 10^{p-1} + ...10 + 1 = \dfrac{10^p - 1}{9}$ donc :
$9x_p + 2 = 10^p + 1$ et
$(9x_p + 2)(x_p) = 10^p x_p + x_p = \dfrac{10^{2p} - 10^p}{9} + \dfrac{10^p - 1}{9} = \dfrac{10^{2p} - 1}{9} = x_{2p}$

1.3.2 Trouver un exemple du même type

On remarque que pour obtenir le même résultat il suffit que :
$a = k * x_p + 1$ et $b = h * x_p + 1$ avec $0 \leq h < k \leq 9$.
On a $a - b = (k - h)x_p$, il faut donc faire en sorte que $a + b = 10^p + 1$
On a $a + b = (h + k)x_p + 2$ c'est à dire prendre $h + k = 9$ avec $k > h$.
Si $k = 8$ et $h = 1$ on a $88888889^2 - 11111112^2 = 7777777777777777$
Si $k = 7$ et $h = 2$ on a $77777778^2 - 22222223^2 = 5555555555555555$
Si $k = 6$ et $h = 3$ on a $66666667^2 - 33333334^2 = 3333333333333333$
Si $k = 5$ et $h = 4$ on a l'exemple traité.

1.4 Montrer que $1...1^2$ (avec k 1 ($k = 1..9$ est un nombre palindrome

On a :
$1^2 = 1$
$11^2 = 121$ (on a $(10 + 1)^2 = 10^2 + 2 * 10 + 1$)
$111^2 = (11 * 10 + 1)^2 = 121 * 10^2 + 22 * 10 + 1 = 12321$

$1111^2 = 12321 * 10^2 + 222 * 10 + 1 = (12321 + 22) * 10^2 + 21 = 1234321$
$11111^2 = (1234321 + 222) * 10^2 + 21 = 123454321$
... $111111111^2 = 12345678987654321$ Mais :
$1111111111^2 = (12345678987654321{+}22222222){*}10^2{+}21 = 1234567900987654321$
n'est pas un palindrome !

1.5 Pourquoi 142857 est-il magique ?

On calculera $142857 * k$ pour $k = 1..6$ On tape :
`(142857*k)$(k=1..6)`
On obtient :
`142857,285714,428571,571428,714285,857142`
On remarque que les nombres obtenus sont formés par les 6 permutations de [1,4,2,8,5,7].

1.6 Si $b = 11115555$ alors $b + 1$ est un carré

1.6.1 Vérification avec `Xcas`

On tape :
`sqrt(11115556)`
On trouve :
`3334`
On vérifie :
$\mathtt{3334}^2 = \mathtt{11115556}$

1.6.2 Généralisation

On remarque que :
$\mathtt{15} + \mathtt{1} = \mathtt{4}^2$
$\mathtt{1155} + \mathtt{1} = \mathtt{34}^2$
Soit $b = 11...155...5$ dont l'écriture contient p fois le chiffre 1 et p fois le chiffre 5. On a donc :
si x s'écrit avec p fois le chiffre 1, on a $x = 11...1 = 10^{p-1} + ...10 + 1 = \frac{10^p-1}{9}$,
$b = x * 10^p + 5x = x(10^p + 5) = \frac{(10^p-1)(10^p+5)}{9} = \frac{(10^{2p}+4*10^p-5)}{9}$
Remarque : Pour ne pas alourdir les notations on ne met pas d'indice à x ni à b.
On en déduit que :
$b + 1 = \dfrac{(10^{2p} + 4 * 10^p + 4)}{9} = (\dfrac{10^p + 2}{3})^2$
On a :
$\dfrac{10^p + 2}{3} = 3(\dfrac{10^p - 1}{9}) + 1 = 3x + 1$
Ou encore :
puisque $10^p = 9x+1$, on a $b+1 = x*10^p+5x+1 = 9x^2+x+5x+1 = (3x+1)^2$
Conclusion :
$b + 1 = 11...155...5 + 1$ est le carré de $33...3 + 1$ dont l'écriture contient p fois le chiffre 3.

1.6.3 Trouver un exemple du même type

On peut faire des essais avec b ayant deux chiffres puis, avec b ayant quatre chiffres : quand $b+1$ est-il un carré ?
Ou bien :
On suppose que $b = m * x * 10^p + n * x$ ($0 < m < 10$ et $0 \leq n < 10$) et on cherche m et n pour que $b+1$ soit un carré.
On a $10^p = 9x+1$ donc $b+1 = 9m * x^2 + m * x + n * x + 1 = (3 * k * x + 1)^2$
On choisit pour m un carré ($m = k^2$) : $m = 1$ ou $m = 4$ ou $m = 9$.
Si $m = 1$ on doit avoir ($m + n = 2 * 3 * k$) $m + n = 6$ et on retrouve l'exemple traité. Si $m = 4$ on doit avoir ($m+n = 2*3*k$) $m+n = 12$ et on trouve $n = 8$, l'exemple cherché est donc : 44...488...8 $+ 1$ est le carré de 66..6 $+ 1$
Si $m = 9$ on doit avoir $m + n = 2 * 3 * 3 = 18$ et on trouve $n = 9$ l'exemple cherché est donc : 99...9 dont l'écriture contient $2p$ fois le chiffre 9 est un carré. Cela on le savait ! ! ! puisque 99...9 $+ 1 = 10^{2p} = (10^p)^2$

1.7 495 et 6174

Étant donné un nombre n de s chiffres au plus non tous égaux, on définit le nombre nA obtenu en rangeant les chiffres de n dans l'ordre croissant et le nombre nD obtenu en rangeant les s chiffres de n dans l'ordre décroissant en rajoutant des zéros pour que nD ait s chiffres (si $n = 21$, $nA = 12$ et $nD = 2100$).
Soit f la fonction qui a n fait correspondre $nD - nA$. On veut étudier les points fixes de f pour $s = 3$ et pour $s = 4$. On va montrer que pour $s = 3$, f a un point fixe qui est égal à 495 et que pour tout n, $f@6(n) = 495$ ($f@@6(n)$ désigne $f(f(f(f(f(f(n))))))$).
On va montrer que pour $s = 4$, f a un point fixe qui est égal à 6174 et que pour tout n, $f@7(n) = 6174$ ($f@@7(n)$ désigne $f(f(f(f(f(f(f(n)))))))$).
On va montrer cela à l'aide d'un programme qui va tester tous les nombres n de 3 (resp 4) chiffres au plus, non tous égaux : ce programme suppose que le résultat est vrai car sinon le programme ne s'arrête pas ! ! ! !
On remarque que :
`f(495)=954-459=495` `f(6174)=7641-1467=6174`

1.7.1 Les chiffres d'un nombre

La procédure `chiffres0` renvoie un couple composé de la liste des chiffres du nombre et du nombre de chiffres.

```
chiffres0(n):={
local s,ch;
ch:=string(n);
s:=size(ch);
return (asc(ch)-[seq(48,s)],s);
}
```

On tape : `chiffres0(6174)` On obtient : `[6,1,7,4],4` On fait la procédure `nombre` qui reconstitue le nombre à partir de la liste de ces chiffres. Il ne faut pas que la chaîne commence par zéro car sinon `expr` considère que la chaîne est une écriture en base 8 (par exemple `expr("016")=14` car en base 8, 14 s'écrit "16").

```
nombre(ch):={
local s,j,chaine;
s:=size(ch);
chaine:=char(ch+[seq(48,s)]);
tantque (chaine[j]==0)faire j:=j+1; ftantque;
chaine:=mid(chaine,j);
return expr(chaine);
}
```

On peut aussi utiliser la commande `horner` de `Xcas` qui donne la valeur en un point d'un polynôme définit par la liste de ses coefficients par puissances décroissantes.
Ainsi `horner([6,1,7,4],10)=6174`

1.7.2 La fonction `f`

On écrit la fonction `etape0` qui a 2 arguments le nombre `n` et le nombre de chiffres autorisés `s` : ainsi `f(n)=etape0(n,4)`

```
etape0(n,s):={
local ch,chA,chD,t,nA,nD;
si (n==0) return n;fsi;
ch,t:=chiffres0(n);
chA:=SortA(ch);
chD:=SortD(ch);
nA:=nombre(chA);
nD:=nombre(chD)*10^(s-t);
return nD-nA;
};
```

En utilisant la fonction `horner` de `Xcas` au lieu de `nombre`, on écrit la fonction `etape` (`f(n)=etape(n,4)`) puis, on écrit la fonction `test` pour tester un nombre quelconque en itérant la fonction `f` :

```
chiffres(n):={
local ch,t;
ch:=string(n);
t:=size(ch);
return (asc(ch)-[seq(48,t)],t);
};
etape(n,s):={
local ch,chA,chD,t,nA,nD;
(ch,t):=chiffres(n);
chA:=SortA(ch);
chD:=SortD(ch);
nA:=horner(chA,10);
nD:=horner(chD,10)*10^(s-t);
```

```
return nD-nA;
};
test(n,s):={
si (n==0) "0 non permis";fsi;
si (n>10^s) alors return s+" chiffres au plus";fsi;
si (irem(n,horner([seq(1,s)],10))==0) alors
return s+" chiffres non identiques"
fsi;
si (s==3) alors
tantque (n!=495) faire print(n);n:=etape(n,s);ftantque;
fsi;
si (s==4) alors
tantque (n!=6174) faire print(n);n:=etape(n,s);ftantque;
fsi;
return n;
};
```

On écrit une première "démonstration" en testant tous les nombres en ordonnant leur chiffres pour ne pas faire 10^3 (resp 10^4) tests : c'est le fait que le programme s'arrête qui prouvera que 495 (resp 6174) est un point fixe de f. On compte le nombre m d'itérations que l'on doit faire avant d'obtenir 495 (resp 6174).
Pour 495 :

```
demo495():={
local j,k,l,u,n,p,m,n0;
p:=0
for (j:=0;j<10;j++){
 for (k:=j;k<10;k++){
  for (l:=k;l<10;l++){
   if (l!=j){
    n:=horner([j,k,l],10);
    p:=p+1;m:=0;n0:=n;
    tantque (n!=495) faire
     m:=m+1;n:=etape(n,3);
    ftantque;
    print(n0,m);
   }
  }
 }
}
return p;
}
```

Pour 6174

```
demo6174():={
local j,k,l,u,n,p,m,n0;
p:=0
for (j:=0;j<10;j++){
for (k:=j;k<10;k++){
```

```
for (l:=k;l<10;l++){
for (u:=l;u<10;u++){
if (u!=j){
n:=horner([j,k,l,u],10);
p:=p+1;m:=0;n0:=n;
tantque (n!=6174) faire  m:=m+1;n:=etape(n,4);ftantque;
print(n0,m);
}
}
}
}
}
return p;
}
```

On trouve que pour $s = 3$, on a tester `p=210` nombres et pour tous ces nombres le nombre d'itérations est inférieur ou égal à 6.
On trouve que pour $s = 4$, on a tester `p=705` nombres et pour tous ces nombres le nombre d'itérations est inférieur ou égal à 7.
On peut par exemple retrouver le nombre `p=705` en comptant :
Les nombres qui ont 4 chiffres différents :
il y en a `comb(10,4)=210` (choix de 4 éléments parmi 10),
Les nombres qui ont 2 chiffres égaux et 2 chiffres différents :
il y en a `comb(3,1)*comb(10,3)=360` (choix de 3 éléments parmi 10 et choix de 1 parmi 3 pour savoir le chiffre qui sera répété),
Les nombres qui ont 2 fois 2 chiffres égaux :
il y en a `comb(10,2)=45` (choix de 2 éléments parmi 10),
Les nombres qui ont 3 chiffres égaux :
il y en a `comb(2,1)3*comb(10,2)=90` (choix de 2 éléments parmi 10 et choix de 1 parmi 2 pour savoir le chiffre qui sera répété),
Donc en tout : `210+360+45+90=705` On remarque que :

— si $s = 3$, $nD = a * 10^2 + b * 10 + c$ avec $a \geq b \geq c$ et $a > c$ on a $nA = c * 10^2 + b * 10 + a$ et $nD - nA = (a - c) * (10^2 - 1)$ ou encore $nD - nA = 9 * (a - c) * 11$ avec $10 > a - c > 0$
  On peut donc faire une démonstration plus rapide puisqu'il suffit de regarder 9 nombres au lieu de 210. Les nombres de la forme $k * (10^2 - 1)$ s'écrivent si $k + k1 = 10$, $[k00] - [k] = [(k - 1)9k1]$, il suffit donc de tester les 5 nombres :
  `099,198,297,396,495.`
— si $s = 4$, $nD = a * 10^3 + b * 10^2 + c * 10 + d$ avec $a \geq b \geq c \geq d$ et $a > d$ on a $nA = d * 10^3 + c * 10^2 + b * 10 + a$ et $nD - nA = (a - d) * (10^3 - 1) + (b - c) * 10^2 - 10$ ou encore $nD - nA = 9 * ((a - d) * 111 + (b - c) * 10)$ avec $10 > a - d > 0$ et $10 > b - c \geq 0$.
  On peut donc faire une démonstration plus rapide puisqu'il suffit de regarder 90 nombres au lieu de 705.
  On écrit :

```
demorapide():={
local j,k,n,p,m,n0;
```

```
for (j:=1;j<10;j++){
for (k:=0;k<10;k++){
n:=9*(j*111+k*10)
m:=0;n0:=n;
tantque (n!=6174) faire  m:=m+1;n:=etape(n,4);ftantque;
print(n0,m);
}
}
return "fin";
}
```

On cherche la liste des nombres à tester :

```
atester():={
local j,k,n,ch,nA,chA,l;
l:=[];
for (j:=1;j<10;j++){
for (k:=0;k<10;k++){
n:=9*(j*111+k*10);
ch:=chiffres(n)[0];
chA:=SortA(ch);
nA:=horner(chA,10);
if (member(nA,l)==0) {l:=append(l,nA);}
}
}
return sort(l);
}
```

On obtient une liste de 30 éléments :

```
[189,288,378,468,558,999,1179,1269,1278,1359,1377,1449,
1467,1557,1899,2268,2358,2367,2448,2466,2556,2799,3357,
3447,3456,3555,3699,4446,4455,4599]
```

On teste les nombres de cette liste dans `demorapide2` :

```
demorapide2():={
local l,s,n,n0,m,j;
l:=atester();
s:=size(l);
for (j:=0;j<s;j++){
n:=l[j];
n0:=n;
m:=0;
tantque (n!=6174) faire  m:=m+1;n:=etape(n,4);ftantque;
print(n0,m);
}
return "fin";
}
```

On trouve par exemple que pour 9351=9*(9*111+4*10) il faut 6 itérations.
Mais comme f(9730)=9730-379=9351, pour 9730 il faut 7 itérations.

1.7.3 Un peu de mathématiques

Comment sécrivent, en base 10, les nombres de la forme `9*(a*111+b*10)` où `a` et `b` sont des entiers entre 0 et 9 avec `0<a`. On a (si on écrit le nombre $100*a+b*10+c$ de chiffres a, b, c sous la forme $[abc]$) :
`9*(a*111+b*10)=(10-1)*(a*111+b*10)=[aaa0-(aa0+a)+b00-b0]=[ab00-ba]=`
si `b>0 [a(b-1)b1a1]` avec `a+a1=10` et `b+b1=9` (cela fait 25 nombres à traiter).
si `b=0 [(a-1)99a1]` avec `a+a1=10` (cela fait 5 nombres à traiter).
Les couples `aa1` et `(b-1)b1` sont donc :
`55 46 37 28 19` et `44 35 26 17 08` Les nombres à traiter sont donc :
`5544 5535 5526 5517 5508 4644 4635 4626 4617 4608 3744 3735`
`3726`
`3717 3708 2844 2835 2826 2817 2808 1944 1935 1926 1917 1908`
et
`999 1899 2799 3699 4599`
On retrouve la liste obtenue précédemment : `[189,288,378,468,558,999,1179,1269,1278,1359,13`
`1467,1557,1899,2268,2358,2367,2448,2466,2556,2799,3357,`
`3447,3456,3555,3699,4446,4455,4599]`

1.7.4 Prolongements

On peut regarder ce qui se passe pour $s = 2$, $s = 5$ etc ... Par exemple pour $s = 2$ les itérés de `12` sont :
`9,81,63,27,45,9....` il n'y a donc pas de point fixe.
On trouve par exemple que les itérés de `02345` sont :
`51975,81972,85932,74943` et `74943` est un point fixe et
que les itérés de `12345` sont `41976...` et `41976` est aussi un point fixe.
que les itérés `31777` sont `63954...` et on trouve un autre point fixe `63954` Il y a donc plusieurs points fixes. Il faut trouver ces points fixes et aussi savoir si les itérés aboutissent toujours a un point fixe ou si il existe des cycles.
Les points fixes trouvés s'écrivent $[ab9(8-b)(10-a)]$, testons alors des nombres de cette formes :
`31977` : on trouve un autre point fixe `83952`
`32967` : on trouve un autre point fixe `73953`
`12969` : on trouve un autre point fixe `86922`
A vous de jouer ! ! ! !

1.8 La suite 1, 2, 2, 3, 3, 3, 4, 4, 4, 4, 5,....

On considère la suite u définie par :
u_0=1 (un 1), $u_1 = u_2$=2 (deux 2), $u_3 = u_4 = u_5$=3 (trois 3) etc...
Quel est la valeur de u_{2010} ?

1.8.1 La correction avec un programme `Xcas`

Attention ! ! ! en `Xcas` les indices commencent à 0.
On peut écrire un programme qui renvoie la liste des $p+1$ premiers termes de cette

suite et qui donne la valeur du dernier terme qui est d'indice p.
On tape :

```
valeurL(p):={
local j,n,L;
j:=0;
n:=1;
L:=[1];
tantque j<p faire
  n:=n+1;
  j:=j+n;
  L:=append(L,n$n);
ftantque;
L:=mid(L,0,p+1);
retourne L,n;
}
:;
```

On tape :
`valeurL(2010)[1]`
On obtient :
`63`

1.8.2 La correction mathématique avec `Xcas`

On a :

- 1 a pour indice 0=1-1
- 2 a pour indice 1 et 2=1+1
- 3 a pour indice 3=1+2, 4 et 5=1+2+(3-1)
- 4 a pour indice 6=1+2+3, 7, 8 et 9=1+2+3+(4-1)
- ....
- p a pour indice $1+2+...+p-1 = p(p-1)/2$,... $p(p-1)/2+p-1 = p(p+1)/2-1$ ou encore si $1+2+...+p-1 = p(p-1)/2 \leq k < p(p-1)/2+p-1 = p(p+1)/2$ alors $u_k = p$

On cherche $p = u_{2010}$ c'est à dire on cherche p vérifiant :
$p(p-1)/2 \leq 2010 < p(p+1)/2$ ou encore
$p(p-1) \leq 4020 < p(p+1)$ ou encore
$p^2-p-4020 \leq 0$ et $p^2+p-4020 > 0$ et $p > 0$
Donc p est entre les racines de $x^2-x-4020$ est est supérieur à la plus grande racine de $x^2+x-4020$ c'est à dire :
$-1/2+\sqrt{1+4*4020}/2 < p \leq 1/2+\sqrt{1+4*4020}/2 < p+1$ c'est à dire
p est la partie entière de $1/2+\sqrt{1+4*4020}/2 = 1/2+\sqrt{1/4+4020}$.
On peut donc utiliser la fonction `round`, on a `round(a)=floor(a+0.5)` et donc `k=round(sqrt(2*p+0.25))`.
On peut remarquer que :
$k(k-1) = (k-1/2)^2-1/4$
$k(k+1) = (k+1/2)^2-1/4$
On cherche k tel que : $k(k-1) = (k-1/2)^2-1/4 \leq 2p < (k+1/2)^2-1/4 = k(k+1)$ c'est à dire $(k-1/2)^2 \leq 2p+1/4 < (k+1/2)^2$ donc `k=round(sqrt(2*p+0.25))` On tape :

```
valeur(p):={
local k;
k:=round(sqrt(2*p+0.25));
retourne k;
}
:;
```

On tape :
`valeur(2010)`
On obtient :
`63`
On tape : `valeur(63*31)`
On obtient : `63`
On tape : `valeur(63*32-1)`
On obtient : `63`
On tape : `valeur(63*32)`
On obtient : `64`
car $63*62/2 = 63*31 = 1953 < 2010 < 63*32-1 = 2015 = 63*62/2+63-1$.
Les 63 termes d'indices 1953,1954,....2015 valent donc 63.

1.9 7 a un multiple qui ne s'écrit qu'avec des 1

1.9.1 Essais avec Xcas

On tape :
`irem(1,7)` réponse `1`
`irem(11,7)` réponse `4`
`irem(111,7)` réponse `6`
`irem(1111,7)` réponse `5`
`irem(11111,7)` réponse `2`
`irem(111111,7)` réponse `0`

1.9.2 Observations

1/ On n'a pas obtenu `3` comme reste car :
`irem(31,7)=3` et donc
`irem(311...1,7)=3`
2/ Pourquoi obtient-on `0` comme reste ?
notons $x_p = 11...1 = \dfrac{10^p-1}{9}$ le nombre s'écrivant avec p fois le chiffre 1. On veut montrer qu'il existe p tel que x_p est divisible par 7. La suite des restes est périodique car les restes sont en nombres finis et supposons que :
$\mathtt{irem}(\mathtt{x_p},7) = \mathtt{irem}(\mathtt{x_q},7)$ avec $p > q$ on a alors :
$\mathtt{irem}(\mathtt{x_p}-\mathtt{x_q},7) = 0$
On a $x_p - x_q$ est divisible par 7 et
$x_p - x_q = \frac{10^p-10^q}{9} = 10^q\frac{10^{p-q}-1}{9} = 10^q x_{p-q}$
Donc $10^q x_{p-q}$ est divisible par 7, comme 10 et 7 sont premiers entre eux on en déduit que x_{p-q} est divisible par 7.

1.9.3 x_{2003} et x_{2004} sont-ils des multiples de 7 ?

On a $x_6 = 111111 = 7 * 15873$ (iquo(111111,7)=15873 et irem(111111,7)=0) donc x_{12} est divisible par 7 ainsi que x_{6k}.
Comme 2004 est divisible par 6, x_{2004} est divisible par 7, par contre 2003=6k+5 donc le reste de la division de x_{2003} par 7 est le même que celui de la division de x_5 par 7 c'est à dire ce reste est égal à 2.

1.10 Tout nombre a un multiple qui ne s'écrit qu'avec des 1 et des 0

1.10.1 Tout nombre premier avec 10 a un multiple qui ne s'écrit qu'avec des 1

Un nombre qui ne s'écrit qu'avec des 1, sera dit nombre en 1, si il s'écrit avec p uns, on le notera x_p. On va tout d'abord montrer que tout nombre n premier avec 10 a un multiple m qui ne s'écrit qu'avec des 1 (il existe k telque $m = x_p = 1..1 = k * n$).
Si n n'est pas premier avec 10, on écrit $n = 2^a * 5^b * q$ (avec pgcd(q,10)=1), on multiplie ce nombre n par 2^{b-a} si $b > a$ ou par 5^{a-b} si $b < a$ pour obtenir le nombre $q * 10^{|a-b|}$ et on applique le résultat précédent à q et obtenir ainsi un multiple de n qui s'écrit avec des 1 suivi par $|a - b|$ zéros. Exemple :
si $n = 37$ on a $37 * 3 = 111$
si $n = 74 = 2 * 37$ on a $74 * 3 * 5 = 74 * 15 = 1110$
si $n = 185 = 5 * 37$ on a $185 * 3 * 2 = 185 * 6 = 1110$

1.10.2 Des remarques

Un nombre qui ne s'écrit qu'avec des 1 sera dit nombre en 1, si il s'écrit avec p uns, on le notera x_p et on a donc :
$x_p = (10^p - 1)/9 = iquo(10^p - 1), 9)$.
Si un nombre premier avec 10 est le diviseur d'un nombre en 1, il divise une infinité de nombre en 1. En effet x_{2p}.... x_{kp} sont des multiple de x_p car on a $x_{2p} = x_p * (10^p + 1)$ et $x_{kp} = x_p * (10^{(k-1)p} + ... + 10^p + 1)$.
On a $x_2 = 11$, $x_3 = 111 = 3 * 37$, $x_4 = 11 * 101$, $x_5 = 11111 = 41 * 271$....
Existe-t-il des nombres en 1 qui soit premiers ? Oui ! il y x_{19} et x_{23}.
On tape :
`isprime(iquo(10^19-1,9))` et on obtient `true`
on tape :
`isprime(iquo(10^23-1,9))` et on obtient `true`
Existe-t-il une infinité de nombres en 1 qui soit premiers ? On ne sait pas ! Étant donné un nombre premier a, on va essayer, dans la suite, de trouver le plus petit p pour que a soit un diviseur de x_p.
Si $a = 3$, on a $p = 3$
Si $a = 37$, on a $p = 3$ et 3 est un diviseur de 37-1
Si $a = 7$, on a $p = 6$ et 6 est un diviseur de 7-1

1.10.3 La suite des restes de 111...1 par n

Pour avoir la suite des restes, on écrit, avec Xcas, le programme suivant :

```
resteun(n):={
local r,q,a,b;
a:=0;
b:=0;
while (irem(n,2)==0) {
n:=iquo(n,2);
a:=a+1;
}
while (irem(n,5)==0) {
n:=iquo(n,5);
b:=b+1;
}
r:=1;print(r);
q:=0;
while (r!=0){
r:=10*r+1;
q:=10*q+iquo(r,n);
r:=irem(r,n);
print(r);
}
if (a>b) q:=q*5^(a-b); else q:=q*2^(b-a);
return(q);
};
```

Ce programme suppose que le résultat est vrai car sinon le programme ne s'arrête pas !!!!
Le fait de faire afficher les restes de la division de 1, 11, 111,... par n montre déjà que la suite des restes est périodique puisque ces restes sont en nombre fini car ils sont positifs ou nuls et inférieurs à n. Mais pourquoi la suite des restes contient-elle toujours 0 ?
Supposons que $(10^k - 1)/9$ et $(10^l - 1)/9$ aient le même reste dans la division par n (avec $pgcd(n, 10) = 1$ et $k > l$). Cela veut dire que $(10^k - 10^l)/9 = 10^l * (10^{k-l} - 1)/9$ est divisible par n donc que $(10^{k-l} - 1)/9$ est divisible par n (n divise $10^l * (10^{k-l} - 1)/9$ n est premier avec 10^l donc n divise $(10^{k-l} - 1)/9$, ce qui prouve bien que la suite des restes contient 0 (le nombre formé par $k - l$ 1 est divisible par n).

1.10.4 Relation entre n et p le nombre de 1 de x_p=111...1 où x_p est un multiple de n

On cherche la fonction R de n qui détermine p, où p est le plus petit entier tel que $x_p = 11...1 = \frac{10^p-1}{9}$ soit un multiple de n. Ce qui veut dire que $\frac{10^p-1}{9} = 0 \mod n$

Quelques essais

Tout d'abord quelques essais :
$x_2 = 11$ est divisible par 11 ($R(11) = 2$)
x_{22} est divisible par 11^2 ($R(11^2) = 22 = 2 * 11$)
$x_3 = 111$ est divisible par 3 et par 37 ($R(3) = 3$ et $R(37) = 3$))
$x_9 = 111111111$ est divisible par 9
x_{27} est divisible par $27 = 3^3$ ($R(3^3) = 3^3$ car $x_{27} = 111111111(10^18 + 10^9 + 1)$
et $10^18 + 10^9 + 1$ est divisible par 3)
x_6 est divisible par 7 ($R(7) = 6$)
x_{42} est divisible par $49 = 7^2$ ($R(7^2) = 42 = 6 * 7$)
x_6 est divisibble par 13 ($R(13) = 3$ car $x_6 = 3 * 7 * 11 * 13 * 37$)
x_6 est divisible par $21 = 3 * 7$ ($R(3 * 7) = 6$)
On va montrer que :
0/ si $p = R(n)$ (x_p est un multiple de n avec p le plus petit possible) alors quelque soit k, x_{k*p} est encore un multiple de n, et réciproquement si x_c est un multiple de n, il existe k tel que $c = k * p$
1/ si n est un multiple de 3 alors x_n est divisible par n ($p = n$)
2/ si n est premier avec 30 alors p est un diviseur de $n - 1$
3/ si n est premier avec 10, si $n = n_1 * n_2$ avec n_1 et n_2 premiers ($n_1 \neq n_2$), et si x_{p_1} est divisible par n_1 et x_{p_2} est divisible par n_2 alors si $p = ppcm(p_1, p_2)$, x_p est divisible par n.
4/ si $n = n_1^k$ alors si x_{p_1} est divisible par n_1 et si $p = p_1 * n_1^{k-1}$ alors x_p est divisible par n.

n est une puissance de 3

On suppose que $n = 3^k$, et on montre par récurrence sur k que $R(3^k) = 3^k$ et que $R(3^k)$ n'est pas un multiple de 3^{k+1}.
Pour $k = 1$, on a $x_1 = 1$ et $x_2 = 11$ ne sont pas divisibles par 3 et $x_3 = 111$ est divisible par 3 et x_3 n'est pas divisible par 9.
supposons la propriété vraie pour k-1, $R(3^{k-1}) = 3^{k-1}$ Posons $b = R(3k - 1) = 3^{k-1}$ et $c = R(3^k)$. Par hypothèse de récurrence x_b est divisible par b et x_c est divisible par $3 * b$ donc par b. Ainsi, d'après la réciproque de la propriété 0, c est un multiple de $R(b) = b$: $c = q * b$
En mettant x_b en facteur dans x_c, on a :
$x_c = x_b(10^{(q-1)b} + ... + 10^b + 1)$
x_b est divisible par 3^{k-1} mais pas par 3^k
x_c est divisible par 3^k donc $10^{(q-1)b} + ... + +10^b + 1$ est divisible par 3, donc comme $10^m = 1 \mod 3$, on en déduit que $q = 0 \mod 3$, et comme c est le plus petit possible $q = 3$ et $c = q * b = 3 * b$.
Donc $x_c = x_b(10^{2*b} + 10^b + 1)$ est divisible par 3^k mais pas par 3^{k+1} car $10^{2*b} + 10^b + 1$ n'est pas divisible par 3 ($10^m = 1 \mod 9$ donc $10^{2*b} + 10^b + 1 = 3 \mod 9$)

n est premier supérieur à 6

On cherche donc p tel que $10^p = 1 \mod n$. La suite des restes possibles est n mais comme n et 9 sont premiers entre eux, il existe u et v avec $0 < v < n$ uniques (identité de Bézout) tels que :
$u * n - v * 9 = 1$ ou encore $10 * v + 1 = u * n + v$ ce qui veut dire que le reste égal à v n'est pas obtenu.
Parmi les $n-1$ restes possibles de la division d'un x_k par n, considérons la relation déquivalence sur les n entiers 0,1,...,n-1 :
$r_1 \simeq r_2$ si il existe k tel que $r_1 * 10^k + x_k = r_2 + q * n$.
On a alors puisque $9 * x_k + 1 = 10^k$, $r_1 - r_2 + (9 * r_1 + 1) * x_k = q * n$.
On a donc p éléments équivalents à 1, Cherchons la periodicité de la suite des restes de $r_1 * 10^k + x_k$ par n, c'est à dire le nombre l d'éléments de la classe r_1. On a $r_1 - r_1 + (9*r_1+1)*x_l = (9*r_1+1)*x_l = q*n$, donc n divise $(9*r_1+1)*x_l$, n est premier avec $(9 * r_1 + 1)$ donc n divise x_l donc $l = p * l_1$.
Mais $r_1 * 10^p + x_p = r_1 \mod n$ donc $l = p$.
Donc si il y a c classes, il y une classe ayant un seul élément et les autres classes ont p éléments donc $n = 1 + p * (c - 1)$ c'est à dire p divise $n - 1$.

1.10.5 Les nombres premiers qui ne s'écrivent qu'avec des 1

Comment sont-ils répartis ?
Un premier test pour trouver le premier nombre premier après 11 qui ne s'écrit qu'avec des 1.
On tape :

```
testun():={
local j;
j:=111;
tantque !isprime(j) faire
j:=10*j+1;
ftantque;
retourne j;
};
```

On tape :
`testun()`
On obtient le nombre qui s'écrit avec avec 19 fois le chiffre 1 :
`1111111111111111111`
Un deuxième test pour trouver le deuxième nombre premier après 11 qui ne s'écrit qu'avec des 1.
On tape :

```
testdeux():={
local j;
j:=testun();
j:=10*j+1;
tantque !isprime(j) faire
j:=10*j+1;
ftantque;
```

```
retourne j;
}:;
```

On tape :
`testdeux()`
On obtient le nombre qui s'écrit avec avec 23 fois le chiffre 1 :
`11111111111111111111111`
Pour avoir les n premiers, on tape :

```
testn(n):={
local j,k,L;
j:=111;
k:=1;
L:=11;
tantque k<n faire
si isprime(j) alors
L:=L,j;
k:=k+1;
fsi;
j:=10*j+1;
ftantque;
retourne L;
}:;
```

On tape :
`testn(4)`
On obtient une liste de 4 nombres et le dernier s'écrit avec avec 317 fois le chiffre 1 :

```
1111111111111111111111111111111111111111111111111111111111111111111111
1111111111111111111111111111111111111111111111111111111111111111111111
1111111111111111111111111111111111111111111111111111111111111111111111
1111111111111111111111111111111111111111111111111111111111111111111111
111111111111111111111111111111111
```

puis `testn(5)` depasse les capacités de `Xcas`

1.11 Le réseau des entiers

Exercice Soient 5 points distincts appartenant au réseau des entiers et les segments définis par ces 5 points.
Combien y-a-t-il de segments ?
Montrer que le milieu d'au moins un segment appartient aussi au réseau.

Il y a `comb(5,2)=10` segments.
Les points ont des abscisses paires ou impaires et des ordonnées paires ou impaires : cela fait 4 sortes de points. Comme on a 5 points, il y a toujours deux points par exemple A et B qui sont de la même sorte.
Si les deux points ont comme coordonnées :

- $A = (2n, 2p)$ et $B = (2m, 2q)$ le milieu M de AB a comme coordonnées : $n + m, p + q$

— $A = (2n, 2p + 1)$ et $B = (2m, 2q + 1)$ le milieu M de AB a comme coordonnèes :
  $n + m, p + q + 1$
— $A = (2n + 1, 2p)$ et $B = (2m + 1, 2q)$ le milieu M de AB a comme coordonnèes :
  $n + m + 1, p + q$
— $A = (2n + 1, 2p + 1)$ et $B = (2m + 1, 2q + 1)$ le milieu M de AB a comme coordonnèes :
  $n + m + 1, p + q + 1$

Donc dans tous les cas M appartient au réseau.

1.12 Le quadrillage

1.12.1 L'énoncé

On considère un rectangle de dimension p, q avec p et q entiers.
On munit ce rectangle d'un quadrillage à coordonnées entières.

1. Écrire une fonction qui trace un rectangle de dimension p, q, son quadrillage et une de ses diagonales.
2. On trace une diagonale de ce rectangle.
   Combien de de carreaux cette diagonale traverse-t-elle ?
   On traitera les exemples : $p = 3, q = 5$, $p = 3, q = 6$,$p = 6, q = 9$,$p = 6, q = 10$,,$p = 6, q = 11$....
   puis on traitera le cas général.

3. Écrire une fonction `nbcarreaux` qui étant donné p, q renvoie le nombre de carreaux traversés par cette diagonale.
   Tracer le nuage des points `(x,y=nbcarreaux(240,x))` pour $0 \geq x \geq 300$. Tracer sur un autre graphique la ligne polygonale reliant ces points.

1.12.2 La solution

1. ```
   quadrillage(p,q):={
    local k,j,L:=segment(0,p+i*q);
   pour j de 0 jusque p faire
   L:=L,segment(j,j+i*q);
   fpour;
   pour k de 0 jusque q faire
   L:=L,segment(i*k,p+i*k);
   fpour
   L;
   }:;
   ```
   On tape :
   `quadrillage(3,5)`
   ```

On obtient :



2. On remarque que la diagonale entre dans le premier carreau, puis elle entre dans un nouveau carreau lorsqu'elle coupe une ligne verticale ou une ligne horizontale ou à la fois une ligne verticale et une ligne vhorizontale.
   Puisqu'il y a $p-1$ verticales et $q-1$ horizontales à traverser, si la diagonale ne coupe jamais à la fois une verticale et une horizontale (c'est à dire si elle ne contient pas de points à coordonnées entières à part le point de départ et le point d'arrivèe) le nommbre de carreaux traversés est $1+p-1+q-1=p+q-1$.
   Si la diagonale coupe r fois, une verticale et une horizontale en même temps, c'est à dire si elle contient $r+2$ points à coordonnées entières (+2 en comptant le point de départ et le point d'arrivèe) le nommbre de carreaux traversés est $p+q-1-r$.
   Que vaut r ?
   La diagonale a comme équation $y=q*x/p=q1*x/p1$ où $p=p1*d$ et $q=q1*d$ avec $d=$pgcd(p,q) et elle aura des points à coordonnées entières chaque fois que x est entier et que $p1$ divise $q1*x$. Puisque $p1$ et $q1$ sont premiers entre eux, $p1$ divise $q1*x$ si x est un entier multiple de $p1$. Cela se produit lorsque $0\geq x\geq p$, pour $x=0, x=p1, x=2*p1 ... x=d*p1=p$, soit $d+1$ fois.
   On a donc $r+2=d+1$ et le nommbre de carreaux traversés est $p+q-$pgcd(p,q).

3. On tape la fonction `nbcarreaux` :

```
nbcarreaux(p,q):={
local d;
d:=gcd(p,q);
return p+q-d;
}
```

On tape :
`nuage_points([x,nbcarreaux(240,x)]$(x=0..300))`
On obtient :



On tape :
`plotlist([x,nbcarreaux(240,x)]$(x=0..300))`
On obtient :



1.13 Résolution dans $\mathbb{N}^*$ de $x^2 = y^2 + z^2$

On veut résoudre dans $\mathbb{N}^*$: $x^2 = y^2 + z^2$.
Par exemple on veut avoir toutes les solutions de $x^2 = y^2 + z^2$ pour $x \leq 200$ et $(x, y, z) \in \mathbb{N}^3$.
On peut écrire un programme qui fait un balayage, mais cela n'est pas efficace car beaucoup trop long !
On tape :

```
triplet0(n):={
  local j,k,s,L;
  L:=NULL;
  pour j de 2 jusque n faire
    pour k de 1 jusque j-1 faire
```

```
      s:=sqrt(j^2-k^2);
      si (type(s)==DOM_INT)alors
        L:=L,[j,k,s],[j,s,k];
      fsi;
    fpour;
  fpour;
  return L;
}
:;
```

Puis :

`A:=triplet0(200):;size(A)`

On obtient (Evaluation time : 2.46) :

`Done, 254`

Il faut donc améliorer le programme.

Cette amélioration donne lieu à l'exercice suivant :

Exercice : amélioration

— Montrer que si (x, y, z) est une solution tous les multiples de (x, y, z) sont aussi des solutions.
En effet :
$(kx)^2 = k^2x^2 = k^2y^2 + k^2z^2 = (ky)^2 + (kz)^2$ pour tout $k \in \mathbb{N}^*$.
On va donc d'abord chercher les solutions pour lesquelles (x, y, z) sont premiers entre eux.

— Montrer que si (x, y, z) sont premiers entre eux dans leur ensemble et vérifient $x^2 = y^2 + z^2$ alors (x, y) (resp (x, z) ou (y, z)) sont premiers entre eux.
En effet :
soit un diviseur premier d de x et de y (resp de x et de z ou de y et de z) alors d^2 divise $x^2 - y^2 = z^2$ (resp $x^2 - z^2 = y^2$ ou $y^2 + z^2 = x^2$) donc d divise z (resp d divise y ou d divise x).

— Montrer que x est impair et que y ou z est divisible par 4.
En effet :
si x est pair x^2 est un multiple de 4 donc $x^2 = 0 \bmod 4$ et
si x est impair x^2 est un multiple de 4 plus 1 donc $x^2 = 1 \bmod 4$.
Supposons que y et z soient impairs :
on a $y^2 = 1 \bmod 4$ et $z^2 = 1 \bmod 4$ donc $y^2 + z^2 = 2 \bmod 4$ donc $y^2 + z^2$ ne peut pas être un carré.
Donc y ou z est pair.

— Montrer que si z est pair alors z est un multiple de 4.
En effet si y est impair et $z = 2 \bmod 4$ alors $y^2 + z^2 = 3 \bmod 4$ donc $y^2 + z^2$ ne peut pas être un carré.

— On pose pour a et b quelconques dans $\mathbb{N}^*$ $(a > b)$:
$x = a^2 + b^2$,
$y = a^2 - b^2$
$z = 2ab$
Montrer qu'alors on a $x^2 = y^2 + z^2$.
En effet :
$x^2 = a^4 + 2a^2b^2 + b^4 = a^4 - 2a^2b^2 + b^4 + 4a^2b^2 = y^2 + z^2$
donc (x, y, z) est une solution dans $\mathbb{N}^*$ de $x^2 - y^2 = z^2$.

— Montrer que toutes les solutions de $x^2 = y^2 + z^2$ avec des nombres entiers premiers entre eux dans leur ensemble sont de cette forme.
  Soit x, y, z une solution de $x^2 = y^2 + z^2$ où x, y, z sont des nombres entiers premiers entre eux dans leur ensemble. On a montrer que y ou z était un multiple de 4. Supposons que ce soit z.
  On a :
  $(\frac{z}{2})^2 = (\frac{x+y}{2})(\frac{x-y}{2})$ et
  puisque x, y, z sont des nombres entiers premiers entre eux dans leur ensemble x et y sont impairs et premiers entre eux.
  Donc $\frac{x+y}{2}$ et $\frac{x-y}{2}$ sont des nombres entiers qui sont premiers entre eux et comme leur produit est un carré, ils sont eux aussi des carrés.
  On pose $a^2 = \frac{x+y}{2}$ et $b^2 = \frac{x-y}{2}$ et on a :
  $x = a^2 + b^2$
  $y = a^2 - b^2$
  $z = 2 * a * b$

On tape :

```
triplet(n):={
local a,b,a2,b2,m,q,p,k,L;
L:=NULL;
for (a:=2;a<sqrt(n);a++) {
  a2:=a^2;
  for (b:=1;b<=sqrt(n-a2) and b<a ;b++) {
    b2:=b^2;
    m:=a2+b2;p:=a2-b2;q:=2*a*b;
    if (gcd(m,p,q)==1){
      for (k:=1;k<=n/m;k++) {
        L:=L,[k*m,k*p,q*k],[k*m,k*q,p*k];
      }
    }
  }
}
return L;
}:;
```

Puis :
`A:=triplet(200):;size(A)`
On obtient instantanément :
`Done, 254`
Dans `A` on a a des triplets comme `[143,55,132]` et `[143,132,55]` qui sont en fait `11*[13,5,12]`, on peut donc refaire un pogramme qui renverra les triplets `[x,y,z]` qui vérifient $x^2 = y^2 + z^2$ avec $y > z$ et `gcd(y,z)=1`. On tape pour avoir les triplets x, y, z vérifiant $x^2 = y^2 + z^2$ avec $x = a^2 + b^2 < n$:

```
triplets(n):={
local a,b,a2,b2,m,q,p,k,L;
L:=NULL;
for (a:=2;a<sqrt(n);a++) {
```

```
  a2:=a^2;
  for (b:=1;b<=sqrt(n-a2) and b<a ;b++) {
    b2:=b^2;
    m:=a2+b2;p:=a2-b2;q:=2*a*b;
    if (gcd(m,p,q)==1){
      L:=L,[m,max(p,q),min(p,q)];
      }
  }
}
return L;
}:;
```

On suppose que n est le nombre de triplets désirés.

```
tripletss(n):={
local a,b,a2,b2,m,q,p,k,L;
L:=NULL;
k:=0;
a:=2;
while (k<n) {
  a2:=a^2;
  for (b:=1;b<a and k<n;b++) {
    b2:=b^2;
    m:=a2+b2;p:=a2-b2;q:=2*a*b;
    if (gcd(m,p,q)==1){
      L:=L,[m,max(p,q),min(p,q)];
      k:=k+1;
    }
  }
a:=a+1;
}
return L;
}:;
```

1.13.1 Exercice

Trouver les solutions en nombres entiers de $x^2 = y^2 + z^2$ avec $x = y + 1$.

1.14 Les paires carrées

1.14.1 L'énoncé

Définition
On dit que les entiers p et q est une paire carrée si il existe deux entiers a er b tels que $q + p = a^2$ et $q - p = b^2$.
Par exemple (6,10) est une paire carrée car $10 - 6 = 2^2$ et $10 + 6 = 4^2$.

Remarque si p et q est une paire carrée alors $2q = a^2 + b^2$ et $2p = a^2 - b^2$ donc $a - b$ est pair et $(2q)^2 = (2p)^2 + (2ab)^2 = a^4 + b^4 - 2a^2b^2 + 4a^2b^2$.
Donc $q^2 = p^2 + (ab)^2$

1. Écrire un programme qui en balayant tous les nombres de 0 à n donne les paires carrées (p, q) avec $0 \leq p \leq q \leq n$,
2. Montrer que si (p, q) est une paire carrée alors on a :
$$2q = a^2 + b^2 \text{ et } 2p = a^2 - b^2$$
3. Montrer que quelque soit $n \in \mathbb{N}$ on soit $n^2 = 1 \bmod 4$, soit $n^2 = 0 \bmod 4$.
En déduire alors que p est pair si (p, q) est une paire carrée.
Modifier votre programme pour tenir compte de cette information.
4. Montrer que $a^2 - b^2$ est un multiple de 4. En déduire que a et b ont même parité et que $a - b$ est pair.
Écrire un programme qui à partir de b et de $a = b + 2n$ calcule les valeurs de p et q vérifiant $q = (a^2 + b^2)/2$ et $p = (a^2 - b^2)/2$ et $0 \leq p \leq q \leq 1000$.
5. Afficher les points de coordonnées (p, q) $(0 \leq p \leq q \leq 1000)$ où (p, q) est une paire carrée.
6. Trouver les équations des droites et des courbes en forme de filets reliants certains de ces points.

1.14.2 La solution

1. ```
paire_carre0(n):={
local a,b,q,p,L;
L:=NULL;
pour p de 0 jusque n faire
pour q de p jusque n faire
a:=sqrt(q+p);
b:=sqrt(q-p);
si (a==floor(a) et b==floor(b)) alors
L:=L,[p,q];
fsi
fpour
fpour
return L
}:;
```
ou on utilise `type` pour tester si `a` et `b` sont des carrés :
```
paire_carre1(n):={
local a,b,q,p,L;
L:=NULL;
pour p de 0 jusque n faire
pour q de p jusque n faire
a:=sqrt(q+p);
b:=sqrt(q-p);
si (type(a)==DOM_INT et type(b)==DOM_INT) alors
L:=L,[p,q];
fsi
fpour
fpour
```
```

```
return L
}:;
```

On tape :
`L1:=paire_carre1(100)`
On obtient (Evaluation time : 2.26) :

```
[0,0],[0,1],[0,4],[0,9],[0,16],[0,25],[0,36],[0,49],
[0,64],[0,81],[0,100],[2,2],[4,5],[6,10],[8,8],[8,17],
[10,26],[12,13],[12,37],[14,50],[16,20],[16,65],
[18,18],[18,82],[20,29],[24,25],[24,40],[28,53],
[30,34],[32,32],[32,68],[36,45],[36,85],[40,41],[42,58],
[48,52],[48,73],[50,50],[54,90],[56,65],[60,61],[64,80],
[70,74],[72,72],[72,97],[80,89],[84,85],[96,100],[98,98]
```

On tape : `dim(L1)`
On obtient ; `49`

2. Si (p, q) est une paire carrée, on a :
   $q + p = a^2$ et $q - p = b^2$
   donc $p = (a^2 - b^2)/2$, donc $a^2 - b^2$ est pair c'est à dire $a^2 - b^2 = 0 \bmod 4$,
   soit $a^2 - b^2 = 2 \bmod 4$.
3. Soit $n \in \mathbb{N}$;
   si n est pair alors $n^2 = 0 \bmod 4$, en effet, si $n = 2k$ on a $n^2 = 4k^2$ et
   si n est impair alors $n^2 = 1 \bmod 4$ en effet :
   si $n = 2k + 1$ on a $n^2 = 4k^2 + 4k + 1$.
4. Donc on a :
   soit $a^2 - b^2 = 0 \bmod 4$,
   soit $a^2 - b^2 = 1 \bmod 4$
   soit $a^2 - b^2 = 3 \bmod 4$.
   et puisque $a^2 - b^2 = 0$ est pair, on en déduit qu $a^2 - b^2 = 0 \bmod 4$ on en déduit que p est pair.
   On tape :

```
paire_carre2(n):={
local a,b,q,p,L;
L:=NULL;
pour p de 0 jusque n pas 2 faire
pour q de p jusque n faire
a:=sqrt(q+p);
b:=sqrt(q-p);
si (a==floor(a) et b==floor(b)) alors
L:=L,[p,q];
fsi
fpour
fpour
return L
}:;
```

On tape :
`L2:=paire_carre2()`
On obtient comme précédement (Evaluation time : 1.36) :

```
[0,0],[0,1],[0,4],[0,9],[0,16],[0,25],[0,36],[0,49],
[0,64],[0,81],[0,100],[2,2],[4,5],[6,10],[8,8],[8,17],
[10,26],[12,13],[12,37],[14,50],[16,20],[16,65],
[18,18],[18,82],[20,29],[24,25],[24,40],[28,53],
[30,34],[32,32],[32,68],[36,45],[36,85],[40,41],[42,58],
[48,52],[48,73],[50,50],[54,90],[56,65],[60,61],[64,80],
[70,74],[72,72],[72,97],[80,89],[84,85],[96,100],[98,98]
```

On tape : `dim(L2)`
On obtient ; `49`

5. On a si (p, q) est une paire carrée : $2p = a^2 - b^2$ donc $a^2 - b^2$ est un multiple de 2 donc
a^2 et b^2 ont même parité donc
a et b ont même parité et donc $a - b$ est un multiple de 2 c'est à dire $a - b$ est pair.
On pose :
$a = b + 2r$ donc $a^2 = b^2 + 4rb + 4r^2$ et on a :
$p = (a^2 - b^2)/2 = 2rb + 2r^2$ et
$q = (a^2 + b^2)/2 = b^2 + 2rb + 2r^2 = b^2 + p$.
On veut obtenir toutes les paires carrées (p, q) vérifiant $0 \geq p \geq q \geq n$,
donc on doit avoir :
$0 \geq b^2 \geq n$ i.e $0 \geq b \geq \sqrt{n}$ et
$0 \geq 2rb + 2r^2 \geq n$ et $0 \geq b^2 + 2rb + 2r^2 \geq n$.
On tape :
`supposons(b>=0 and b<=31);`
`simplify(solve(b^2+2*r*b+2*r^2<1000,r))`
On obtient ; :
`[((r>((-b-sqrt(-b^2+2000))/2)) && (((-b+sqrt(-b^2+2000))/2)>r))]`
On tape :

```
paire_carre(n):={
local b,q,p,L,r,nmax;
L:=NULL;
pour b de 0 jusque sqrt(n) faire
nmax:=(sqrt(-b^2+2*n))/2-b/2;
pour r de 0 jusque nmax faire
p:=2*r*b+2*r^2;
q:=b^2+p;
L:=L,[p,q];
fpour
fpour
return L
}:;
```

Puis, on tape :
`L:=paire_carre(100):;`
Le calcul est tres rapide ! ! ! ! !
On tape : `dim(L)`
On obtient instantanément : `48`
On tape :

`L3:=paire_carre(1000):;`
Le calcul est tres rapide ! ! ! ! !
On tape : `dim(L3)`
On obtient : `421`

6. On tape : `nuage_points(L3)`
   On obtient :



7. On a $q = p + b^2$ donc les points sont sur les droites d'équations $y = x + b^2$ pour b fixé.
   On a $p = 2nb + 2n^2$ donc $b = (p/(2n) - n)$ donc $q = p^2/(4n^2) + n^2$ donc les points sont sur les courbes d'équations $y = x^2/(4n^2) + n^2$ pour n fixé.
   On tape et on obtient :



1.15 Les triangles rectangles presque isocèles

Définition

1. Un triangle rectangle est presque isocèle de type 1 si ses côtés sont les entiers a, $c - 1$, c où c est la longueur de l'hypoténuse.
2. Un triangle rectangle est presque isocèle de type 2 si ses côtés sont les entiers a,$a + 1$,c où c est la longueur de l'hypoténuse.

Pour trouver les triangles rectangles presque isocèles il faut et il suffit de résoudre en nombre entiers les équations

1. $a^2 + (c - 1)^2 = c^2$, c'est à dire :
   trouver les couples $(a, c) \in \mathbb{N}$ vérifiant $a^2 + c^2 - 2c + 1 = c^2$.
2. $a^2 + (a + 1)^2 = c^2$, c'est à dire :
   trouver les couples $(a, c) \in \mathbb{N}$ vérifiant $2a^2 + 2a + 1 = c^2$.

1.15.1 L'énoncé 1

On veut trouver les triangles rectangles presque isocèle de type 1.
On cherche donc les couples $(a, c) \in \mathbb{N}$ vérifiant $a^2 + c^2 - 2c + 1 = c^2$ c'est à dire : $a^2 = 2c - 1$.

1.15.2 La solution de l'énoncé 1

a est donc impair i.e. $a = 2k + 1$ avec $k \in \mathbb{N}$.
On a donc $4k^2 + 4k = 2(c - 1)$ et donc $c = 2k^2 + 2k + 1 = k^2 + (k + 1)^2$
Les côtés du triangle rectangle presque isocèle de type 1 sont donc :
$[2k + 1, k^2 + (k + 1)^2 - 1, k^2 + (k + 1)^2]$ ($k \in \mathbb{N}$) ou encore
$[2k + 1, 2k(k + 1), k^2 + (k + 1)^2]$ ($k \in \mathbb{N}$).
On tape :

```
[2k+1,2k*(k+1),k^2+(k+1)^2]$(k=0..10)
```

On obtient :

```
[1,0,1],[3,4,5],[5,12,13],[7,24,25],[9,40,41],[11,60,61],
[13,84,85],[15,112,113],[17,144,145],[19,180,181],[21,220,221]
```

On peut trouver une relation de récurrence entre 2 triplets successifs.
On cherche x, y, z tels que pour tout $k \in \mathbb{N}$ on ait :
$a_k = xa_{k-1} + yb_{k-1} + zc_{k-1}$.
$2k + 1 = x * (2k - 1) + y * 2k * (k - 1) + z * (k^2 + (k - 1)^2 = 2k^2(z + y) + 2k(x - y - z) + z - x$
Donc $y = -z$ $x - y - z = x = 1$ et $z - x = 1$
On trouve donc $x = 1, y = -2, z = 2$
On cherche x, y, z tels que :
$b_k = xa_{k-1} + yb_{k-1} + zc_{k-1}$.
On trouve $x = 2, y = -1, z = 2$
On cherche x, y, z tels que :
$c_k = xa_{k-1} + yb_{k-1} + zc_{k-1}$.
On trouve $x = 2, y = -2, z = 3$
On a donc :

```
A:=[[1,-2,2],[2,-1,2],[2,-2,3]]
```

Le $k-$ième triplet est égal à $A^k * [1, 0, 1]$
On tape :

```
(A^k*[1,0,1])$(k=1..6)
```

On obtient :

```
[3,4,5],[5,12,13],[7,24,25],[9,40,41],[11,60,61],[13,84,85],
[15,112,113],[17,144,145],[19,180,181],[21,220,221]
P,B:=jordan(A)
```

On obtient :

```
[[0,2,0],[4,2,0],[4,2,1]],[[1,1,0],[0,1,1],[0,0,1]]
```

On a $A^k = PB^kP^{-1}$
On sait que $B^n = [[1, n, \sum_{k=1}^{n-1} k], [0, 1, n], [0, 0, 1]]$ donc on tape :

```
Bn:=unapply([[1,n,sum(k,k=1..n-1)],[0,1,n],[0,0,1]],n)
P*Bn(5)*inv(P)*[1,0,1]
```

On obtient :

```
[11,60,61]
```

On tape :

```
subst([2k+1,2k*(k+1),k^2+(k+1)^2],k,5)
```
On obtient :
```
[11,60,61]
```

1.15.3 L'énoncé 2

On veut trouver les triangles rectangles presque isocèle de type 2.
On cherche donc les couples $(a, c) \in \mathbb{N}$ vérifiant $2a^2 + 2a + 1 = c^2$.

1. Écrire un programme qui donne, en balayant tous les entiers a de 0 à n, les triplets $(a, a+1, c) \in \mathbb{N}^3$ vérifiant $a^2 + (a+1)^2 = c^2$.
   Donner les solutions pour $n = 1000$.
2. Écrire un programme qui donne les n premiers triplés en modifiant le programme `tripletss` (cf 1.13).
3. Montrer que si le triplet $(a, a+1, c) \in \mathbb{N}^3$ vérifie $a^2 + (a+1)^2 = c^2$ alors c est impair et vérifie $2c^2 - 1 = (2a+1)^2$.
   Écrire un programme qui donne les triplets $(a, a+1, c) \in \mathbb{N}^3$ vérifiant $a^2 + (a+1)^2 = c^2$ et qui utilise cette proriété de c.
4. On considère la suite récurrente :
   $a_0 = 0,\ a_1 = 3,\ a_{n+2} = 6a_{n+1} - a_n + 2\ \ (n \geq 0)$
   Montrer par récurrence que pour tout $n \in \mathbb{N}$ on a :
   $a_{n+1}^2 - 6a_n a_{n+1} + a_n^2 - 2a_{n+1} - 2a_n = 3$
5. On considère la suite récurrente :
   $c_0 = 1,\ c_1 = 5,\ c_{n+2} = 6c_{n+1} - c_n\ \ (n \geq 0)$
   Montrer par récurrence que pour tout $n \in \mathbb{N}$ on a :
   $c_n^2 = 2a_n^2 + 2a_n + 1$
   En déduire que pour tout $n \in \mathbb{N}$, les triplets $(a_n, a_n + 1, c_n)$ donnent des solutions.
   On montrera dans la question 9 que l'on obtient ainsi toutes les solutions.
6. Écrire un programme qui donne des triplets $(a, a+1, c) \in \mathbb{N}^3$ vérifiant $a^2 + (a+1)^2 = c^2$ et qui utilise pour c la relation de récurrence :
   $c_0 = 1,\ c_1 = 5,\ c_{n+2} = 6c_{n+1} - c_n\ \ (n \geq 0)$.
7. On veut trouver c_n en fonction de n.
   Déterminer les progressions géométriques $v_n = v_0 * r^n$ qui vérifie la relation de $v_{n+2} = 6v_{n+1} - v_n$.
   Puis en déduire la valeur de c_n en fonction de n.
8. On veut trouver a_n en fonction de n.
   Déterminer p pour que la suite $u_n = a_n + p$ vérifient la relation de récurrence $u_{n+2} = 6u_{n+1} - u_n$.
   Puis en déduire la valeur de a_n en fonction de n
9. Montrer que l'on a : $c_0 = 1, a_0 = 0$ et
   $c_{n+1} = 4a_n + 3c_n + 2$
   $a_{n+1} = 3a_n + 2c_n + 1$
   et réciproquement si c_n et a_n vérifient :
   $c_0 = 1,\ a_0 = 0,$ et
   $c_{n+1} = 4a_n + 3c_n + 2$
   $a_{n+1} = 3a_n + 2c_n + 1$

alors
$c_0 = 1,\ c_1 = 5,\ c_{n+2} = 6c_{n+1} - c_n\ \ (n \geq 0)$ et
$a_0 = 0,\ a_1 = 3,\ a_{n+2} = 6a_{n+1} - a_n + 2\ \ (n \geq 0)$

10. Montrer que l'on obtient toutes les solutions cherchèes
11. Écrire la relation $c_{n+2} = 6c_{n+1} - c_n$ sous la forme :
$$\begin{pmatrix} c_{n+2} \\ c_{n+1} \end{pmatrix} = M \begin{pmatrix} c_{n+1} \\ c_n \end{pmatrix}$$
et déterminer la matrice M
12. Calculer par récurrence M^n, pour $n \in \mathbb{N}$.
13. On considère les suites w définies par $w_0 \in \mathbb{N},\ w_{n+1}$ =floor$((3+2\sqrt{2})w_n)$ (floor désigne la partie entière)
Montrer qu'alors pour tout $n \in \mathbb{N}^*$ on a :
$w_{n+1} = 6w_n - w_{n-1}$
En déduire que si on choisit $w_0 = 1$ la suite w_n est la suite c_n qui donne la longueur des hypoténuse des triangles rectangles presque isocèles.
En déduire que si $w_0 = 1$ alors $2 * w_n^2 - 1$ est le carré de $2a_n + 1$.
Écrire un programme qui donne les triplets $(a, a+1, c) \in \mathbb{N}^3$ vérifiant $a^2 + (a+1)^2 = c^2$ et qui utilise cette définition de c.
14. Écrire les relations de récurrence :
$c_n = 4a_{n-1} + 3c_{n-1} + 2$ et $a_n = 3a_{n-1} + 2c_{n-1} + 1$
avec une matrice 3x3.
15. On considère les suites v définies par :
$v_0 \in \mathbb{N},\ v_{n+1} = 3+$`floor`$((3+2\sqrt{2})v_n)$ (`floor` désigne la partie entière)
Montrer qu'alors pour tout $n \in \mathbb{N}^*$ on a :
$v_{n+1} = 6v_n - v_{n-1} + 2$
En déduire que si on choisit $v_0 = 0$ la suite v_n est la suite a_n qui donne la longueur du plus petit coté des triangles rectangles presque isocèles.
En déduire que si $v_0 = 0$ et $w_0 = 1$ alors $2 * w_n^2 - 1 = (2v_n + 1)^2$.
Écrire un programme qui donne les triplets $(a, a+1, c) \in \mathbb{N}^3$ vérifiant $a^2 + (a+1)^2 = c^2$ et qui utilise cette définition de a et c.

1.15.4 La solution de l'énoncé 2

1. On tape :

```
rectpiso1(n):={
local a,c,d,L;
L:=NULL;
pour a de 0 jusque n faire
d:=2a^2+2a+1;
c:=round(sqrt(d));
si c^2==d alors L:=L,[a,a+1,c];fsi;
fpour;
retourne L;
}:;
```

On tape :
`rectpiso1(10000)`
On obtient (Evaluation time : 12.73) :
`[0,1,1],[3,4,5],[20,21,29],[119,120,169],[696,697,985],`
`[4059,4060,5741]`

2. On change le test `if (gcd(m,p,q)==1)` en :
   `if (gcd(m,p,q)==1 and abs(p-q)==1)` et on tape

```
triplets2(n):={
local a,b,a2,b2,m,q,p,k,L;
L:=NULL;
k:=0;
a:=2;
while (k<n) {
  a2:=a^2;
  for (b:=1;b<a and k<n;b++) {
    b2:=b^2;
    m:=a2+b2;p:=a2-b2;q:=2*a*b;
    if (gcd(m,p,q)==1 and abs(p-q)==1){
      L:=L,[m,max(p,q),min(p,q)];
      k:=k+1;
    }
  }
a:=a+1;
}
return L;
}:;
```

   On tape :
   `triplets2(5)`
   On obtient :
   `[5,4,3],[29,21,20],[169,120,119],[985,697,696],[5741,4060,4059]`

3. Puisque $c^2 = a^2 + (a+1)^2 = 2a^2 + 2a + 1$ on en déduit que c^2 est un entier impair donc c est un entier impair.
   On a aussi : $2c^2 - 1 = 4a^2 + 4a + 1 = (2a+1)^2$
   On tape :

```
rectpiso2(n):={
local a,c,d,L;
L:=NULL;
pour c de 1 jusque n pas 2 faire
d:=2c^2-1;
a:=round(sqrt(d));
si a^2==d alors a:=(a-1)/2; L:=L,[a,a+1,c];fsi;
fpour;
retourne L;
}:;
```

   On tape :
   `rectpiso2(10000)`

On obtient (Evaluation time : 5.9) :
`[0,1,1],[3,4,5],[20,21,29],[119,120,169],[696,697,985],`
`[4059,4060,5741]`

4. Pour $n = 0$ on a :
$a_{n+1}^2 - 6a_n a_{n+1} + a_n^2 - 2a_{n+1} - 2a_n = 3^2 - 2 * 3 = 3$
Supposons que pour $n \in \mathbb{N}$ on a :
$a_{n+1}^2 - 6a_n a_{n+1} + a_n^2 - 2a_{n+1} - 2a_n = 3$
Calculons :
$a_{n+2}^2 - 6a_{n+1} a_{n+2} + a_{n+1}^2 - 2a_{n+2} - 2a_{n+1}$
sachant que :
$a_{n+2} = 6a_{n+1} - a_n + 2$
Dans `Xcas`, `a2` désigne a_{n+2},`a1` désigne a_{n+1} et `a0` désigne a_n.
On tape :
`a2:=6a1-a0+2`
`normal(a2^2-6a1*a2+a1^2-2a2-2a1)`
On obtient :
`a0^2-6*a0*a1-2*a0+a1^2-2*a1`
Donc :
$a_{n+2}^2 - 6a_{n+1} a_{n+2} + a_{n+1}^2 - 2a_{n+2} - 2a_{n+1} = a_{n+1}^2 - 6a_n a_{n+1} + a_n^2 - 2a_{n+1} - 2a_n =$
Donc, d'après l'hypothèse de récurrence on a :
$a_{n+2}^2 - 6a_{n+1} a_{n+2} + a_{n+1}^2 - 2a_{n+2} - 2a_{n+1} = 3$
Donc, pour tout $n \in \mathbb{N}$, on a :
$$a_{n+1}^2 - 6a_n a_{n+1} + a_n^2 - 2a_{n+1} - 2a_n = 3$$

5. Pour $n = 0$ on a :
$c_0^2 = 2a_0^2 + 2a_0 + 1 = 1$ (puisque $a_0 = 0$)
Pour $n = 1$ on a :
$c_1^2 = 2a_1^2 + 2a_1 + 1 = 18 + 6 + 7 = 25$ (puisque $a_1 = 3$)
Supposons que pour $n \in \mathbb{N}$ on ait :
$c_n^2 = 2a_n^+ 2a_n + 1$
$c_{n+1}^2 = 2a_{n+1}^2 + 2a_{n+1} + 1$
Dans `Xcas`, `c2` désigne c_{n+2},`c1` désigne c_{n+1} et `c0` désigne c_n.
On tape :
`c2:=6c1-c0`
`normal(c2^2)`
On obtient :
`c0^2-12*c0*c1+36*c1^2`
On a d'après l'hypothèse de récurrence : `c0^2=2a0^2+2a0+1` et
`c1^2=2a1^2+2a0+1` :
Donc si on pose `C=c2^2`, on a :
`C:=normal(2a0^2+2a0+1+36*(2a1^2+2a1+1)-12c0*c1)`
On obtient :
`2*a0^2+2*a0+72*a1^2+72*a1-12*c0*c1+37`
On tape :
`A:=normal(2a2^2+2a2+1)`
On obtient :

`-a0^2+6*a0*a1+2*a0-a1^2+2*a1+3`
On veut montrer que `C=A`.
On tape :
`normal((C-A)/12)`
On obtient :
`2*a0*a1+a0+a1-c0*c1+2`
Il faut donc montrer que :
$c0 * c1 = 2 * a0 * a1 + a0 + a1$ ou encore que :
$c0^2 * c1^2 = (2 * a0 * a1 + a0 + a1 + 2)^2$ ou encore que :
$(2a0^2 + 2a0 + 1) * (2a1^2 + 2a1 + 1) - (2 * a0 * a1 + a0 + a1 + 2)^2 = 0$
On tape :
`normal((2a0^2+2a0+1)*(2a1^2+2a1+1)-(2*a0*a1+a0+a1+2)^2)`
On obtient :
`a0^2-6*a0*a1-2*a0+a1^2-2*a1-3`
qui vaut bien 0 d'apres la question 2.
Donc pour tout $n \in \mathbb{N}$ on a $c_n^2 = 2a_n^2 + 2a_n + 1 = a_n^2 + (a_n + 1)^2$.
Donc le triplet $(a_n, a_n + 1, c_n)$ est une solution pour tout n.
On peut montrer que l'on obtient ainsi toutes les solutions (cf question9).

6. On tape :

```
rectpiso3(n):={
local a,c,d,L,c0,c1;
c0:=1;
c1:=5;
L:=[0,1,1];
tantque c1<n faire
d:=2*c1^2-1;
a:=(sqrt(d)-1)/2;
L:=L,[a,a+1,c1];
c:=c1;
c1:=6*c1-c0;
c0:=c;
ftantque;
retourne L;
}:;
```

On tape :
`rectpiso3(10000)`
On obtient (instantanément) :
`[0,1,1],[3,4,5],[20,21,29],[119,120,169],[696,697,985],`
`[4059,4060,5741]`

7. Si $v_n = v_0 * r^n$ vérifie $v_{n+2} = 6v_{n+1} - v_n$ c'est que r est solution de $x^2 - 6x + 1 = 0$.
On tape :
`r1,r2:=solve(x^2-6x+1,x)`
On obtient :
`[-2*sqrt(2)+3,2*sqrt(2)+3]`
Donc il y a 2 progressions gépmétriques de raison $r1 = -2\sqrt{2} + 3$ et $r2 = 2\sqrt{2} + 3$ qui verifient aussi cette relation de récurence.

Une combinaison lineaire de suites verifiant $v_{n+2} = 6v_{n+1} - v_n$ verifient aussi cette relation de récurence.
Donc la suite :
$w_n = x * r1^n + y * r2^n$ vérifie $w_{n+2} = 6w_{n+1} - w_n$ quelque soit x et y.
Puisque la suite c_n est entièrement déterminée par $c_0 = 1$, $c_1 = 5$ et par la relation de récurence $c_{n+2} = 6c_{n+1} - c_n$, pour avoir $c_n = w_n$ il suffira d'avoir $c_0 = w_0$ et $c_1 = w_1$ c'est à dire :
$x + y = 1$ et $x * r1 + y * r2 = 5$.
On tape :
```
linsolve([x+y=1,x*r1+y*r2=5],[x,y])
```
On obtient :
```
[(-(sqrt(2))+2)/4,(sqrt(2)+2)/4]
```
Donc :

$$c_n = \frac{2-\sqrt{2}}{4} * (-2 * \sqrt{2} + 3)^n + \frac{2+\sqrt{2}}{4} * (2 * \sqrt{2} + 3)^n$$

8. On cherche p pour que la suite $u_n = p + a_n$ vérifient la relation de $u_{n+2} = 6u_{n+1} - u_n$. On sait que a_n vérifie $a_{n+2} = 6a_{n+1} - a_n + 2$ donc p doit vérifier :
$p + a_{n+2} = 6p + 6a_{n+1} - p - a_n$ donc $p + 2 = 6p - p$.
On tape :
```
solve(p+2=6p-p,p)
```
On obtient :
```
[1/2]
```
Donc comme précédemment on va déterminer la suite $a_n + 1/2$.
Pour cela on cherche x et y tel que :
$x + y = a_0 + 1/2 = 1/2$ et $x * r1 + y * r2 = a_1 = 3 + 1/2 = 7/2$
On tape :
```
normal(linsolve([x+y=1/2,x*r1+y*r2=7/2],[x,y]))
```
On obtient :
```
[(-(sqrt(2))+1)/4,(sqrt(2)+1)/4]
```
On a donc :
$u_n = \frac{1-\sqrt{2}}{4} * (-2 * \sqrt{2} + 3)^n + \frac{1+\sqrt{2}}{4} * (2 * \sqrt{2} + 3)^n$
Donc :

$$a_n = -\frac{1}{2} + \frac{1-\sqrt{2}}{4} * (-2 * \sqrt{2} + 3)^n + \frac{1+\sqrt{2}}{4} * (2 * \sqrt{2} + 3)^n$$

On tape :
```
c(n):=((-sqrt(2)+2)/4)*(-2*sqrt(2)+3)^n+
((sqrt(2)+2)/4)*(2*sqrt(2)+3)^n
a(n):=((-sqrt(2)+1)/4)*(-2*sqrt(2)+3)^n+
((sqrt(2)+1)/4)*(2*sqrt(2)+3)^n-1/2
normal(a(5),a(5)+1,c(5))
```
On obtient :
```
4059,4060,5741
```
On tape :
```
normal(a(6),a(6)+1,c(6))
```
On obtient :

`23660,23661,33461`
On tape :
`normal(a(10),a(10)+1,c(10))`
On obtient :
`27304196,27304197,38613965`

9. On suppose que pour tout $n \in \mathbb{N}$ on a :
$c_0 = 1,\ c_1 = 5,\ a_0 = 0,\ a_1 = 3$ et
$c_{n+2} = 6c_{n+1} - c_n$ et
$a_{n+2} = 6a_{n+1} - a_n + 2$
Montrons par récurrence qu'alors pour tout $n \in \mathbb{N}$ on a :
$c_0 = 1,\ a_0 = 0$ et
$c_{n+1} = 4a_n + 3c_n + 2$
$a_{n+1} = 3a_n + 2c_n + 1$ La relation est vrai pour $n = 0$ et $n = 1$ car :
$c_1 = 4 * 0 + 3 * +2 = 5$ et
$a_1 = 3 * 0 + 2 * 1 + 1 = 3$
Si la relation est vrai pour n on a : $c_n = 4a_{n-1} + 3c_{n-1} + 2$
$a_n = 3a_{n-1} + 2c_{n-1} + 1$
alors
$3c_n - c_{n-1} = 12a_{n-1} + 8c_{n-1} + 6 = 4a_n + 2$
et
$3a_n - a_{n-1} = 8a_{n-1} + 6c_{n-1} + 3 = 2c_n - 1$
donc
$c_{n+1} = 3c_n + 3c_n - c_{n-1} = 3c_n + 4a_n + 2$
et
$a_{n+1} = 3a_n + 3a_n - a_{n-1} + 2 = 3a_n + 2c_n + 1$
Réciproquement si pour tout $n \in \mathbb{N}$ on a :
$c_0 = 1,\ a_0 = 0$ et
$c_{n+1} = 4a_n + 3c_n + 2$
$a_{n+1} = 3a_n + 2c_n + 1$
Alors :
$c_0 = 1,\ a_0 = 0$ et $c_1 = 4*0+3*1+2 = 5,\ a_1 = 3*0+2*1+1 = 3$ et
$3c_{n+1} - 4a_{n+1} = c_n + 2$
$2c_{n+1} - 3a_{n+1} = -a_n + 1$
soit
$4a_{n+1} = 3c_{n+1} - c_n - 2$
$2c_{n+1} = 3a_{n+1} - a_n + 1$
et puisque : $c_{n+2} = 4a_{n+1} + 3c_{n+1} + 2$
$a_{n+2} = 3a_{n+1} + 2c_{n+1} + 1$
on en déduit que pour tout $n \in \mathbb{N}$:
$c_{n+2} = 6c_{n+1} - c_n$
$a_{n+2} = 6a_{n+1} - a_n + 2$

10. Supposons que l'on ait $(a+1)^2 + a^2 = 2a^2 + 2a + 1 = c^2$.
Alors c est impair et a et c sont premiers entre eux.
On a aussi : $2c^2 = (2a+1)^2 + 1$
Si a est pair
0n a :
$(a+1)^2 = c^2 - a^2 = (c+a)(c-a)$
donc si d divise $a+1$, d divise soit $c+a$, soit $c-a$ mais d ne divise pas

$c + a$ et $c - a$ car a et c sont premiers entre eux donc :
il existe p et q tel que $c + a = p^2$ et $c - a = p^2$ et $pq = a + 1$ On a donc $p \geq q$ avec p et q sont impairs et,
$2a = p^2 - q^2$ et $2c = p^2 + q^2$.
La relation $4c^2 = 8a^2 + 8a + 4$ donne comme relation entre p et q :
$(p^2 + q^2)^2 = 2(p^2 - q^2)^2 + 4(p^2 - q^2) + 4$
donc
$p^4 + q^4 - 6p^2q^2 + 4p^2 - 4q^2 + 4 = 0$
Posons
$X = p^2$
$Y = q^2$ X et Y sont des carrés d'entiers qui vérifient :
$Y^2 - 2Y(2 + 3X) + (X + 2)^2$
Le discriminant de ce trinôme en Y est donc le carré d'un entier, c'est à dire $2X + 2$ est le carré d'un entier puisque :
$(2 + 3X)^2 - (X + 2)^2 = 8X^2 + 8X = (2p)^2(2X + 2)$.
En posant $p = 2 * p_1 + 1$ on en déduit que $X = 4p_1^2 + 4p_1 + 1$ et que $2X + 2 = 8p_1^2 + 8p_1 + 4 = 4(2p_1^2 + 2p_1 + 1)$ et donc qu'il existe $k \in \mathbb{N}$ tel que :
$2p_1^2 + 2p_1 + 1 = k^2$
et on a alors :
$Y = q^2 = (2 + 3p^2) - 4kp$ (avec le signe "-" car $q < p$)
donc si $2a^2 + 2a + 1 = c^2$ avec a est pair, il existe p avec $p = 2 * p_1 + 1$ et k vérifiant $2p_1^2 + 2p_1 + 1 = k^2$ (ou $2k^2 = p^2 + 1$) tel que :
$2a = p^2 - q^2 = X - Y = 4kp - 2 - 2p^2 = 4kp - 4k^2 = 4k(p - k)$
$2c = p^2+q^2 = X+Y = 4p^2+2-4kp = 8k^2-4kp-2 = 2(4k^2-2kp-1)$
Réciproquent si on a $2p_1^2 + 2p_1 + 1 = k^2$ pour $p_1 \in \mathbb{N}^*$ et $k \in \mathbb{N}$ alors si on pose $p = 2 * p_1 + 1$ on a $2k^2 = (2p_1 + 1)^2 + 1 = p^2 + 1$ et :
$a = 2kp - 1 - p^2 = 2k(p - k)$
$c = 4k^2 - 2kp - 1$
a est pair $a > p_1 > 0$, $c > k$ et on vérifie que $2a^2 + 2a + 1 = c^2$, pour cela, on tape :
```
a:=2k*p-k^2
c:=4k^2-1-2k*p
factor(2a^2+2a+1-c^2)
```
On obtient :
```
2*p^2*(2*k^2-p^2-1)
```
Donc puisque $2k^2 = p^2 + 1$ on a $2a^2 + 2a + 1 = c^2$
Si a est impair alors $a + 1$ est pair
0n fait le même genre de raisonnement :
$a^2 = c^2 - (a + 1)^2 = (c + a + 1)(c - a - 1)$
donc si d divise a, d divise soit $c + a + 1$, soit $c - a - 1$ mais d ne divise pas $c + a + 1$ et $c - a - 1$ car a et c sont premiers entre eux donc :
il existe p et q tel que $c + a + 1 = p^2$ et $c - a - 1 = p^2$ et $pq = a$.
On a $p \geq q$ et p et q sont impairs et,
$2a = p^2 - q^2 - 2$ et $2c = p^2 + q^2$.
Posons
$X = p^2$
$Y = q^2$ X et Y sont des carrés d'entiers qui vérifient :

$X^2 - 2X(2+3Y) + (Y+2)^2$
(même équation en échangeant Y et X)
$2Y+2$ est un carré en posant $q = 2q_1 + 1$ cela veut dire qu'il existe $r \in \mathbb{N}$ tel que : $2q_1^2 + 2q_1 + 1 = r^2$ ou encore $2r^2 = q^2 + 1$
et alors $X = 3Y + 2 + 4qr = 3q^2 + 2 + 4qr$ (avec le signe "+" car $X = p^2 > q^2$) donc :
$2a = p^2 - q^2 - 2 = 2q^2 + 4qr = 2(q^2+1) - 2 + 4qr = 2(2r^2 + 2qr - 1)$
$2c = p^2 + q^2 = 4q^2 + 4qr + 2 = 4(q^2+1) - 2 + 4qr = 2(4r^2 + 2qr - 1)$
Réciproquent si on a $2q_1^2 + 2q_1 + 1 = r^2$ pour $q_1 \in \mathbb{N}$ et $r \in \mathbb{N}$ alors si on pose $q = 2 * q_1 + 1$ on a $2r^2 = (2q_1+1)^2 + 1 = q^2 + 1$ et :
$a = 2r^2 + 2qr - 1$
$c = 4r^2 + 2qr - 1$
a est impair et on tape :

```
a:=2r^2+2q*r-1
c:=4r^2+2q*r-1
factor(2a^2+2a+1-c^2)
```

On obtient :

```
2*q^2*(2*r^2-q^2-1)
```

Donc puisque $2r^2 = q^2 + 1$ on a $2a^2 + 2a + 1 = c^2$.
Conclusion Si a_1, c_1 est une solution de $2a^2 + 2a + 1 = c^2$ alors cette solution engendre 2 nouvelles solutions qui sont :
$a_2 = 2c_1(2a_1 - c_1 + 1)$ et $c_2 = (2a_1+1)^2 + 2c_1(c_1 - 2a_1 - 1)$
$a_3 = (2a_1+1)^2 + 2(2a_1+1)c_1$ et $c_3 = 2(2a_1+1)^2 + 2(2a_1+1)c_1 + 1$
puis la solution a_2, c_2 engendre 2 nouvelles solutions qui sont :
$a_4 = 2c_2(2a_2 + 1 - c_2)$ et $c_4 = (2a_2+1)^2 + 2c_2(c_2 - 2a_2 - 1)$
$a_5 = (2a_2+1)^2 + 2(2a_2+1)c_2$ et $c_5 = 2(2a_2+1)^2 + 2(2a_2+1)c_2 + 1$
On remarquera que la solution $[0, 1]$ engendre $[0, 1]$ et $[3, 5]$, que la solution $s_1 = [3, 5]$ engendre $s_2 = [20, 29]$ et $s_3 = [119, 169]$ etc...par ce processus la solution s_n engendre les solutions s_{2n} et s_{2n+1}, puis la solution s_{n+1} engendre les solutions s_{2n+2} et s_{2n+3} etc...
Pour faire le lien avec les suites précédentes, il reste a montrer que la suite $[a_n, c_n]$ ainsi engendrée vérifie :
$c_n = 4a_{n-1} + 3c_{n-1} + 2$ et $a_n = 3a_{n-1} + 2c_{n-1} + 1$
On utilise `Xcas` pour faire les calculs.
Si $s_n =$`[a,c]` et $s_{n+1} =$`a1,c1)` sont des solutions successives, on désigne s_{2n} par `[sa1(a,c),sc1(a,c)]`, s_{2n+1} par `[sa2(a,c),sc2(a,c)]` et s_{2n+2} par `[sa1(a1,c1),sc1(a1,c1)]`.
On désigne par `rc(a,c)` et par `ra(a,c)` les relations de récurrence qui donne c_{n+1} et a_{n+1} en fonction de a_n et c_n.
On suppose que les relations de récurrence sont vérifiées par s_n et s_{n+1} i.e. que `a1=ra(a,c)` et `c1=rc(a,c)`.
On tape les définitions :

```
sa1(a,c):=normal(4*c*a+2*c-2*c^2)
sc1(a,c):=normal(4*c^2-4*c*a-2*c-1)
sa2(a,c):=normal(4*c*a+2*c-1+2*c^2)
sc2(a,c):=normal(4*c^2+4*c*a+2*c-1)
rc(a,c):=4*a+3*c+2;
ra(a,c):=3*a+2*c+1
```

On tape :
`normal(rc(sa1(a,c),sc1(a,c))-sc2(a,c))`
On obtient : `0`
On tape :
`normal(ra(sa1(a,c),sc1(a,c))-sa2(a,c))`
On obtient : `0`
On tape :
`factor(ra(sa2(a,c),sc2(a,c))-sa1(ra(a,c),rc(a,c)))`
On obtient : `-8*(2*a^2+2*a-c^2+1)`
On tape :
`factor(rc(sa2(a,c),sc2(a,c))-sc1(ra(a,c),rc(a,c)))`
On obtient : `-8*(2*a^2+2*a-c^2+1)`
ce qui donne bien `0` puisque $2a^2 + 2a + 1 = c^2$ car s_n `=[a,c]` est une solution de cette équation par hypothèse.

On a ainsi montrer que si :
s_n `=[a,c]` et s_{n+1} `=a1,c1)` sont des solutions successives qui vérifient les relations de récurrence alors s_{2n}, s_{2n+1}, s_{2n+2} vérifient aussi les relations de récurrence.
$s_1 = [3, 5]$ engendre $s_2 = [20, 29]$ et s_3. On peut vérifier par le programme de la question 1 que $[3, 5]$ et $[20, 29]$ sont 2 solutions successives qui vérifie la relation de récurrence (29=4*3+3*5+2 et 20=3*3+2*5+1) donc on en déduit par récurrence que : si s_2 engendre $s_4$et s_5, alors s_2, s_3, s_4 sont des solutions successives qui vérifient la relation de récurrence.... On peut aussi faire le programme suivant qui renvoie la liste des solutions en utilisant les fonctions `sa1, sc1, sa2, sc2` définies précédemment :

```
sa1(a,c):=normal(4*c*a+2*c-2*c^2);
sc1(a,c):=normal(4*c^2-4*c*a-2*c-1);
sa2(a,c):=normal(4*c*a+2*c-1+2*c^2);
sc2(a,c):=normal(4*c^2+4*c*a+2*c-1);
tripisoc(n):={
local a,a1,a2,c,c1,c2,L,k;
L:=[0,1],[3,5];
pour k de 1 jusque n faire
a,c:=L[k];
L:=L,[sa1(a,c),sc1(a,c)],[sa2(a,c),sc2(a,c)];
fpour;
return L;
}:;
```

11. 0n a :
$c_{n+2} = 6c_{n+1} - c_n$ et
$c_{n+1} = c_{n+1}$
donc

$$M = \begin{pmatrix} 6 & -1 \\ 1 & 0 \end{pmatrix}$$

12. Si
$$M^n = \begin{pmatrix} p_n & q_n \\ s_n & t_n \end{pmatrix}$$
On a :
$p_0 = 1$, $q_0 = 0$, $s_0 = 0$, $t_0 = 1$, $p_1 = 6$, $q_1 = -1$, $s_1 = 1$, $t_0 = 0$
$M^{n+1} = M * M^n$ c'est à dire :
$p_{n+1} = 6p_n - s_n$, $q_{n+1} = 6q_n - t_n$, $s_{n+1} = p_n$, $t_{n+1} = q_n$
donc $p_{n+1} = 6p_n - p_{n-1}$ avec $p_0 = 1$ et $p_1 = 6$
$q_{n+1} = 6q_n - q_{n-1}$ avec $q_0 = 0$ et $q_1 = -1$
Les suites p_n et q_n vérifient la même relation de récurrence que c_n donc :
p_n et q_n sont des combinaisons linéaires des 2 progressions géométriques de raison $r1 = -2\sqrt{2} + 3$ et $r2 = 2\sqrt{2} + 3$.
Calcul de $p_n = x(r1)^n + y(r2)^n$
Pour trouver p_n en fonction de n il faut donc résoudre :
$[x + y = 1 = p_0, x * r1 + y * r2 = 6 = p_1]$
On tape :
```
r1,r2:=solve(x^2-6x+1,x)
```
On obtient :
```
[-2*sqrt(2)+3,2*sqrt(2)+3]
```
On tape :
```
linsolve([x+y=1,x*r1+y*r2=6],[x,y])
```
On obtient :
```
[(-3*sqrt(2)+4)/8,(3*sqrt(2)+4)/8]
```
Calcul de $q_n = x(r1)^n + y(r2)^n$
Pour trouver q_n en fonction de n il faut donc résoudre :
$[x + y = 0 = q_0, x * r1 + y * r2 = 1 = q_1]$
On tape :
```
linsolve([x+y=0,x*r1+y*r2=1],[x,y])
```
On obtient :
```
[(-(sqrt(2)))/8,(sqrt(2))/8]
```
Donc pour $n \in \mathbb{N}$:
$p_n = (-3\sqrt{2} + 4)(-2\sqrt{2} + 3)^n + (3\sqrt{2} + 4)(2\sqrt{2} + 3)^n$
$q_n = -\sqrt{2}(-2\sqrt{2} + 3)^n + \sqrt{2}(2\sqrt{2} + 3)^n$

13. On peut aussi écrire la relation de récurrence :
$c_n = 4a_{n-1} + 3c_{n-1} + 2$
$a_n = 3a_{n-1} + 2c_{n-1} + 1$
en utilisant A une matrice 3x3 :
$$A = \begin{pmatrix} 3 & 4 & 2 \\ 2 & 3 & 1 \\ 0 & 0 & 1 \end{pmatrix}$$
et ainsi, on a :
$$\begin{pmatrix} c_{n+1} \\ a_{n+1} \\ 1 \end{pmatrix} = A \begin{pmatrix} c_n \\ a_n \\ 1 \end{pmatrix}$$
On a donc :

$$\begin{pmatrix} c_n \\ a_n \\ 1 \end{pmatrix} = A^n \begin{pmatrix} 1 \\ 0 \\ 1 \end{pmatrix}$$

On vérifie en tapant :

```
normal(rsolve([a(n+1)=3*a(n)+2*c(n)+d(n),
c(n+1)=4*a(n)+3*c(n)+2*d(n),
d(n+1)=d(n)],[a(n),c(n),d(n)],[a(0)=0,c(0)=1,d(0)=1]))
```

On obtient :

```
[[(-(sqrt(2))+1)/4*(-2*sqrt(2)+3)^n+
(sqrt(2)+1)/4*(2*sqrt(2)+3)^n+(-1)/2,
(-(sqrt(2))+2)/4*(-2*sqrt(2)+3)^n+
(sqrt(2)+2)/4*(2*sqrt(2)+3)^n,
1]]
```

14. On considère la suite :
$w_0 \in \mathbb{N}^*,\ w_{n+1}$ =floor$(3+2*\sqrt{2})*w_n)$ (floor désigne la partie entière)
On veut montrer que pour tout $n \in \mathbb{N}^*$ on a $w_{n+1}+w_{n-1}=6w_n$.
On a :
$w_{n+1} \leq (3+2*\sqrt{2})*w_n < w_{n+1}+1$
Donc :
$(3+2*\sqrt{2})*w_n - 1 < w_{n+1} \leq (3+2*\sqrt{2})*w_n$
On a :
$w_n \leq (3+2*\sqrt{2})*w_{n-1} < w_n+1$
Puisque $\dfrac{1}{3+2*\sqrt{2}} = 3-2*\sqrt{2}$ on a :
$w_n*(3-2*\sqrt{2}) \leq w_{n-1}) < (w_n+1)*(3-2*\sqrt{2})$
Donc :
$(3+2*\sqrt{2})*w_n - 1 + w_n*(3-2*\sqrt{2}) < w_{n+1}+w_{n-1} < (3+2*\sqrt{2})*w_n + (w_n+1)*(3-2*\sqrt{2})$
Donc :
$6w_n - 1 < w_{n+1}+w_{n-1} < 6w_n + (3-2*\sqrt{2}) < 6w_n+1.$
Comme $w_{n+1}+w_{n-1}$ est un entier, on en déduit qwe :
$w_{n+1}+w_{n-1} = 6w_n.$
Pour montrer que w_n est la suite c_n il suffit de montrer que $w_1=5$ lorsque $w_0=1$.
On a bien $w_1 = floor(3+2*\sqrt{2}) = 5$, donc w_n et c_n coincident.
On tape :

```
rectpiso5(n):={
local a,c,d,L;
L:=NULL;
c:=1;
tantque c<n faire
d:=2*c^2-1;
a:=(sqrt(d)-1)/2;
L:=L,[a,a+1,c];
c:=floor((3+2*sqrt(2))*c);
ftantque;
```

```
retourne L;
}:;
```

On tape :
`rectpiso4(10000)`
On obtient (instantanément) :
`[0,1,1],[3,4,5],[20,21,29],[119,120,169],[696,697,985],`
`[4059,4060,5741]`

15. On considère la suite :
$v_0 \in \mathbb{N}^*,\ v_{n+1}$ =floor$(3+2*\sqrt{2})*w_n)+3$ (floor désigne la partie entière)
On veut montrer que pour tout $n \in \mathbb{N}^*$ on a $v_{n+1}+v_{n-1}=6w_n+2$.
On a :
$v_{n+1} \leq (3+2*\sqrt{2})*v_n+3 < v_{n+1}+1$
Donc :
$(3+2*\sqrt{2})*v_n+2 < v_{n+1} \leq (3+2*\sqrt{2})*v_n+3$
On a :
$v_n \leq (3+2*\sqrt{2})*v_{n-1}+3 < v_n+1$
Puisque $\dfrac{1}{3+2*\sqrt{2}}=3-2*\sqrt{2}$ on a :
$(v_n-3)*(3-2*\sqrt{2}) \leq v_{n-1}) < (v_n-2)*(3-2*\sqrt{2})$
Donc :
$(3+2*\sqrt{2})*v_n+2+(v_n-3)*(3-2*\sqrt{2}) < v_{n+1}+v_{n-1} < (3+2*\sqrt{2})*v_n+3+(v_n-2)*(3-2*\sqrt{2})$
Donc :
$6v_n+1 < 6v_n+2-9+6*\sqrt{2} < v_{n+1}+v_{n-1} < 6w_n+(3-6+4*\sqrt{2}) < 6v_n+3$.
Comme $v_{n+1}+v_{n-1}$ est un entier, on en déduit qwe :
$v_{n+1}+v_{n-1}=6v_n+2$.
Pour montrer que v_n est la suite a_n il suffit de montrer que $v_1=3$ lorsque $v_0=0$.
On a bien $v_1=floor((3+2*\sqrt{2})*0)+3=3$, donc v_n et a_n coincident.
On tape :

```
rectpiso5(n):={
local a,c,d,L;
L:=NULL;
c:=1;
a:=0;
tantque c<n faire
L:=L,[a,a+1,c];
c:=floor((3+2*sqrt(2))*c);
a:=floor((3+2*sqrt(2))*a)+3;
ftantque;
retourne L;
}:;
```

On tape :
`rectpiso5(10000)`
On obtient (instantanément) :

```
[0,1,1],[3,4,5],[20,21,29],[119,120,169],[696,697,985],
[4059,4060,5741]
```

1.16 La suite a_n est-elle une suite d'entiers ?

1.16.1 La définition de a_n

On définit la suite a_n par :
$a_0 = 1$ et
$a_n = \dfrac{1 + a_0^2 + a_1^2 + a_2^2 + ... + a_{n-1}^2}{n}$ pour $n > 0$
La suite a_n est-elle une suite d'entiers ?

1.16.2 Calcul des premiers termes de a_n

On a :
$a_0 = 1,\ a_1 = (1 + a_0^2)/1 = 2,\ a_3 = (1 + a_0^2 + a_1^2)/2 = 3,\ a_4 = (1 + 1^2 + 2^2 + 3^2)/3 = 5$ On remarque que pour $n > 0$ on a :
$n * a_n = 1 + a_0^2 + a_1^2 + ... + a_{n-1}^2$ On a donc pour $n > 0$:
$a_n = \dfrac{(n-1)a_{n-1} + a_{n-1}^2}{n} = \dfrac{a_{n-1}(n - 1 + a_{n-1})}{n}$ On tape pour calculer les n premiers termes de a_n :

```
lasuite(n):={
  local j,u,r,s,L;
    L:=1,2;
    u:=2; j:=1;
    L:=L,1;
    tantque j<n faire
      u:=j*u+u*u;
      j:=j+1;
      u:=u/j;
      L:=L,u;
   ftantque;
  return [L];
}:;
```

On tape :
`L:=lasuite(10)`
On obtient les 11 premiers termes de la suite :
`[1,2,3,5,10,28,154,3520,1551880,267593772160,7160642690122633501504]`
On tape :
`L[10]`
On obtient :
`7160642690122633501504`
Les 11 premiers termes sont entiers et on remarque qu a_{11} est de l'ordre de $7e+21$ et donc la suite croit très vite !

1.16.3 Calcul des termes de a_n modulo un nombre premier

Après des essais infructueux pour montrer que la suite a_n est une suite d'entiers, on se décide à chercher un contre-exemple ! Puisque $a_0 = 1$ et
$a_n = \dfrac{1 + a_1^2 + a_2^2 + ... + a_{n-1}^2}{n}$ pour $n > 0$
On a : $na_n = 1+a_1^2+a_2^2+...+a_{n-1}^2 = (n-1)a_{n-1}+a_{n-1}^2 = a_{n-1}(n-1+a_{n-1})$.
Donc si pour un nombre p premier on a $a_{p-1}(p-1+a_{p-1})$ est divisible par p c'est à dire si $a_{p-1}(p-1+a_{p-1})\%p == 0\%p$ cela voudra dire que a_p est un entier et si $a_{p-1}(p-1+a_{p-1})\%p! = 0\%p$ cela voudra dire que a_p n'est pas un entier.
Pour cela on fait un programme qui calule pour un nombre premier p les $a_k \bmod p$ pour $k = 0, 1, ..p$ et on regarde si $a_{p-1}(p-1+a_{p-1})$ est divisible par p.
Remarque
On prend p premier pour que lors des calculs des $a =_n$ $(n < p)$, n soit inversible dans $\mathbb{Z}/p\mathbb{Z}$ ($a_n\%p = ((n-1)a_{n-1} + a_{n-1}^2)/n\%p$).
On tape :

```
leterme(p):={
//pour p premier, leterme(p) renvoie
//(p-1)*a(p-1)+a(p-1)*a(p-1) % p (=p*a(p) % p)
local j,u;
    u:=2 % p; j:=1;
    tantque j<p-1 faire
      u:=((j*u+u*u)/(j+1);
      j:=j+1;
    ftantque;
  return (((p-1)*u+u*u) % p);
}:;
```

On tape la liste des nombres premiers inférieurs à 90 :
`P:=[2,3,5,7,11,13,17,19,23,29,31,37,41,43,47,53,59,61,67,71,73,79,83,89]`
On tape :
`L:=NULL;pour p in P do L:=L,leterme(p) fpour;`
On obtient :
`[],[0 % 2,0 % 3,0 % 5,0 % 7,0 % 11,0 % 13,0 % 17,0 % 19,0 % 23,0 % 29,0 % 31,0 % 37,0 % 41,-19 % 43,0 % 47,0 % 53,0 % 59,12 % 61,18 % 67,0 % 71,0 % 73,0 % 79,19 % 83,0 % 89]`
On trouve des termes non nuls :
`-19 % 43, 12 % 61, 18 % 67, 19 % 83`
Cela signifie que a_{43} n'est pas un entier !
Mais cela ne signifie pas que a_{47} est un entier car Les termes de la suite pour $n > 43$ risquent d'être des fractions ayant pour dénominateur des puissances de 43.
a_{61} (resp a_{67}, a_{83}) seront des fractions ayant un dénominateur divisible par 61 (resp 67,83).
On tape une fonction booléenne qui teste si a_n est un entier lorsque $n-1$ est un nombre premier inférieur ou égal à 43 :

```
estentiera(n):={
```

```
local P;
P:=[2,3,5,7,11,13,17,19,23,29,31,37,41,43,47,53,59,61,67,71,73,79,83,89,97];
si member(n, P)  alors
  si leterme(n)!=0 % (n) alors  return faux;
    sinon
      si n<43 alors
        return vrai;
      fsi
  fsi;
fsi;
return "?";
}:;
```

On tape :
`estentiera(23)`
On obtient :
`vrai`
On tape :
`estentiera(43)`
On obtient :
`faux`
On tape :
`estentiera(47)`
On obtient :
`"?"`
`estentiera(33)`
On obtient :
`"?"`

1.16.4 Prolongements

On peut définir le même type de suite en utilisant des cubes à la place des carrés ou encore d'autres puissances à la place des carrés.
Les premières valeurs de ces suites sont aussi des entiers et :
pour le cube a_{89} est non entier.
pour la puissance 4, a_{97} est non entier.
pour la puissance 5, a_{251} est non entier.
pour la puissance 6, a_{19} est non entier.
pour la puissance 7, a_{239} est non entier.
Pour la puissance k, puisque $a_0 = 1$ et
$a_n = \dfrac{1 + a_1^k + a_2^k + ... + a_{n-1}^k}{n}$ pour $n > 0$
On a : $na_n = 1 + a_1^k + a_2^k + ... + a_{n-1}^k = (n-1)a_{n-1} + a_{n-1}^k$.
On tape :

```
lasuitek(n,k):={
  local j,u,r,s,L;
    u:=2; j:=1;L:=1,u;
    tantque j<n faire
```

```
      u:=j*u+u^k;
      j:=j+1;
      u:=u/j;
      L:=L,u;
    ftantque;
  return [L];
}:;

letermek(p,k):={
//pour p premier, letermek(p,k) renvoie
//(p-1)*a(p-1)+a(p-1)^k % p (=p*a(p) % p)
local j,u;
    u:=2 % p; j:=1;
    tantque j<p-1 faire
      u:=j*u+u^k;
      j:=j+1;
      u:=u/j;
    ftantque;
  return (p-1)*u+u^k;
}:;

estentierak(n,k):={
local P,p0,p1,p2,L;
p0:=2,3,5,7,11,13,17,19,23,29,31,37,41,43,47,53,59,61,67,71,73,79,83,8
p1:=101,103,107,109,113,127,131,137,139,149,151;
p2:=157,163,167,173,179,181,191,193,197,199,211,223,227,229,233,239,2
P:=[p0,p1,p2];
L:=[1,1,43,89,97,251,19;239];
si member(n, P)  alors
  si letermek(n,k)!=0 % (n) alors  return faux;
    sinon
      si n<L[k] alors
        return vrai;
      fsi
  fsi;
fsi;
return "?";
}:;
```

On a pris `L:=[1,1,43,89,97,251,19,239]` pour que `n` soit un nombre premier inférieur ou égal à 43 (resp 89,97,251,19,239) lorsque `k=2` (resp 3,4,5,6,7) car on a :

`letermek(89,3)` renvoie `41% 89`
`letermek(97,4)` renvoie `-22% 97`
`letermek(251,5)` renvoie `-83% 251`
`letermek(19,6)` renvoie `-9% 19`
`letermek(239,7)` renvoie `-117% 239`

1.16.5 Remarque

On tape :

```
letermekmod(n,k,p):={
//pour p premier, letermekmod(n,k,p) renvoie
//(n-1)*a(n-1)+a(n-1)^k % p (=n*a(n) % p)
local j,u;
    u:=2 % p; j:=1;
    tantque j<n-1 faire
      u:=(j*u+u^k)/(j+1);
      si u!=(0 % p) alors print(j+1,u);fsi;
      j:=j+1;
    ftantque;
  return (((n-1)*u+u^k)/n) % p;
}:;
```

On tape :
`letermekmod(34,2,2),letermekmod(34,2,17)`
On obtient en vert :
2,3 % 11
3,5 % 11
4,-1 % 11
5,-5 % 11
`0 % 3,0 % 11`
Cela signifie de a_{33} est un entier. On tape :
`letermekmod(33,2,3),letermekmod(33,2,11)`
On obtient en vert :
2,3 % 11
3,5 % 11
4,-1 % 11
5,-5 % 11
`0 % 2,0 % 17`
Cela signifie de a_{34} est un entier.

Chapitre 2

Des Quiz

2.1 Nombres de croisements

Pour relier les villes A et B il y a une seule ligne de bus dont les horaires sont faciles à retenir.
Les bus partent chaque jour de A vers B et de B vers A, toutes les heures de 7h à 19h i.e. à 7h, 8h,.....19h.
Les bus suivent la même route et le trajet dure 5h.
On cherche le nombre de croisements qu'un bus partant de A, croise sur la route de bus provenant de B ? (on ne compte pas comme croisement, un bus qui arrive lorsque l'autre part).
1/ Combien de fois le bus de 7h, partant de A, croise sur la route de bus provenant de B ?
2/ Combien de fois le bus de 12h, partant de A, croise sur la route de bus provenant de B ?
3/ Combien de fois le bus de 17h, partant de A, croise sur la route de bus provenant de B ?
4/ Faire un programme `nbcroisement(t)` qui renvoie le nombre de croisements d'un bus partant de A à t heures avec les bus provenant de B.

Solution 1/Le bus de 7h, partant de A, croise en premier le bus de 7h venant de B : ils se croisent donc milieu du trajet soit au bout de 2h30 soit à 9h30. Le bus de 7h arrive en B à 12h au moment ou le bus de 12h part de B donc :
le bus de 7h partit de A a croisé les bus de 7h, 8h, 9h, 10, 11h en provenance de B il y a donc eu 1+11-7=5 croisements sur la route.
2/Le bus de 12h part de A lorsque le bus de 7h en provenance de B arrive en A. Le bus de 12h partant de A arrive en B à 17h au moment ou le bus de 17h part de B donc :
le bus de 12h partit de A a croisé le bus de 8h, 9h, 10h, 11h, 12h, 13h, 14h, 15, 16h. Il y a donc eu 1+167-8=9 croisements sur la route.
3/Le bus de 17h part de A lorsque le bus de 12h en provenance de B arrive en A. Le bus de 17h partant de A arrive en B à 22h : le dernier bus part de B à 19 h donc :
le bus de 17h partit de A a croisé les bus de 13h, 14h, 15, 16h, 17h, 18h, 19h en provenance de B. Il y a donc eu 19-12=7 croisements sur la route.
4/Si le bus part de A à th, il part lorsque le bus de $t-5$h en provenance de B arrive. Le bus partit de A à th arrive en B à $t+5$h lorsque le bus de $t+5$h. Puisque

$7 \leq t \leq 19$, on a $(t-4) \leq 19$ et $7 \leq (t+4)$, le bus qui part de A à th croise les bus (si ils existent i.e. si $7 \leq (t-4)$ et $(t+4) \leq 19$) de $t-4$h,...$t+4$h en provenance de B. Donc :

— si $7 \leq (t-4)$ et $(t+4) \leq 19$ i.e. si $11 \leq t$ et $t \leq 24$ le bus qui part de A à th croise les bus de $t-4$h...$t+4$h en provenance de B donc il y a 1+$t+4-(t-4)=9$ croisements.
— si $(t-4) \leq 7$ le bus qui part de A à th croise les bus de 7h,...$t+4$h en provenance de B donc il y a $1+t+4-7=t-2$ croisements.
— si $19 \leq (t+4)$ le bus qui part de A à th croise les bus de $t-4$h,...19h en provenance de B donc il y a $1+19-(t-4)=24-t$ croisements.

Le programme par exemple avec des `si` :

```
nbcroisement(t):={
si t<7 ou t>19 alors retourne "erreur" fsi;
si  t>=11 et t<=15 alors retourne 9 fsi;
si  t<11 alors retourne t-2; sinon retourne 24-t; fsi;
}:;
```

On tape :
`nbcroisement(t)$(t=7..19)`
On obtient :
`5,6,7,8,9,9,9,9,9,8,7,6,5`
Ou plus simplement :
`nbcroisements(t):=1+min((t+4),19)-max((t-4),7)` On tape :
`nbcroisements(t)$(t=7..19)`
On obtient :
`5,6,7,8,9,9,9,9,9,8,7,6,5`

2.2 Ordre de grandeur : pliage d'une feuille de papier

Une feuille de papier de format $A3$ (dimension 420 mm $\times$ 297 mm) a une épaisseur de 0.1mm.
On plie cette feuille 7 fois sur elle-même (vous pouvez essayer !).
Qu'elles sont les dimensions du parallélépipède obtenu ?
On plie cette feuille 20 fois sur elle-même.
Si cela etait possible qu'elles seraient les dimensions du parallélépipède obtenu ?
Solution
Soit une feuille d'épaisseur de 0.1mmm et de dimension de $a \times b$ avec a=420mm et b=297mm.
Au 1ier pliage on obtient le format $A4$ dépaisseur de $2 \times 0.1 = 0.2$mm et de dimensions $a/2 \times b$.
Au 2ième pliage on obtient une épaisseur de $2^2 \times 0.1 = 0.4$mm et une dimension de $a/2 \times b/2$.
Au 3ième ($3 = 2 * 2 - 1$) pliage on obtient une épaisseur de $2^3 \times 0.1 = 0.8$mm et une dimension de $a/2^2 \times b/2$.

Au 4ième $(4 = 2 * 2)$ pliage on obtient uneépaisseur $2^4 \times 0.1 = 128$mm=12.8cm et une dimension de $a/2^2 \times b/2^2$.
Au 5ième $(5 = 2 * 3 - 1)$ pliage on obtient une épaisseur de $2^5 \times 0.1 = 32$mm = 3.2cm et une dimension de $a/2^3 \times b/2^2$.
Au 6ième $(6 = 2 * 3)$ pliage on obtient uneépaisseur $2^6 \times 0.1 = 64$mm=6.4cm et une dimension de $a/2^3 \times b/2^3$.
Au 7ième $(7 = 2*4-1)$ pliage on obtient uneépaisseur de $2^7 \times 0.1 = 128$mm=12.8cm et une dimension de $a/2^4 \times b/2^3 = 2.625cm \times 3.7125cm$.
Au 8ième $(8 = 2 * 4)$ pliage on obtient uneépaisseur $2^8 \times 0.1 = 256$mm=25.6cm dépaisseur et une dimension de $a/2^8 \times b/2^4 = 2.625cm \times 1.85625cm$.
....
Au 20ième $(20 = 2*10)$ pliage on obtiendrait uneépaisseur $2^{20} \times 0.1 = 104857.6$mm=104.857m dépaisseur et une dimension de $a/2^{10} \times b/2^{10} = 0.41015625mm \times 0.2900390625mm$.

Chapitre 3

Les injections, les surjections et les bijections

Notation $f@g$ est la composé des applications f et g et
$f@@n$ désigne $fof..of$ (n fois) (ce sont les notations de tt Xcas)

3.1 Existence d'une bijection entre]0,4[et [1,3]

On définit la fonction $g(x) = 1 + x/2$ et pour $x \in]0, 4[$, la fonction $f(x)$ par :
si $x \in]2 - (1/2)^n, 2 - (1/2)^{n+1}[\cup]2 + (1/2)^n, 2 + (1/2)^{n+1}[$ pour $n \in \mathbb{N}$ alors $f(x) = g(x)$ si il existe $n \in \mathbb{N}$ tel que $x = 2 - (1/2)^n$ ou $x = 2 + (1/2)^n$ alors $f(x) = x$ On tape :
`(2-(1/2)^n)$(n=0..5)`
On obtient :
`1,3/2,7/4,15/8,31/16,63/32`
On tape :
`(2+(1/2)^n)$(n=0..5)`
On obtient :
`3,5/2,9/4,17/8,33/16,65/32`
On tace une partie du graphe de f (pour $x \in]0, 7/4] \cup]9/4, 4[$) on tape dans un niveau de géométrie :

```
affichage(plotfunc(x/2+1,x=0..4),4+epaisseur_ligne_2);
segment(0,4,affichage=1+epaisseur_ligne_2);
segment(i,3i,affichage=2+epaisseur_ligne_2);
point(1+i,affichage=4+epaisseur_point_2);
point(1+3*i/2,affichage=point_carre+epaisseur_point_2);
point(3/2+7*i/4,affichage=point_carre+epaisseur_point_2);
point(3/2*(1+i),affichage=4+epaisseur_point_2);
point(7/4*(1+i),affichage=4+epaisseur_point_2);
point(7/4+15*i/8,affichage=point_carre+epaisseur_point_2);
point(3*(1+i),affichage=4+epaisseur_point_2);
point(3+5*i/2,affichage=point_carre+epaisseur_point_2);
point(5/2*(1+i),affichage=4+epaisseur_point_2);
```

```
point(5/2+9*i/4,affichage=point_carre+epaisseur_point_2);
point(9/4*(1+i),affichage=4+epaisseur_point_2);
point(9/4+17*i/8,affichage=point_carre+epaisseur_point_2);
```

On obtient en bleu le graphe de f sur $]0,7/4]\cup]9/4,4[$:



La fonctions f est discontinue en $x = 2 \pm (1/2)^n$. Les points bleus s'accumulent sur le point d'affixe $2 + 2 * i$. Les points noirs n'appartiennent pas au graphe de f alors qu'ils appartiennent au graphe de g.
On remarquera que f est continue au point $x = 2$

3.2 Existence d'une bijection entre $\mathbb{R}$ et [-1,1]

On définit la fonction $g(x) = \text{atanh}(x)$ et pour $x \in \mathbb{R}$, la fonction $f(x)$ par :
si $x \in]-\infty, -1[\cup]\text{atanh}@@n(-1), \text{atanh}@@(n+1)(-1)[\cup]\text{atanh}@@n(1), \text{atanh}@@(n+1)(1)[\cup]1, +\infty[$ pour $n \in \mathbb{N}$ alors $f(x) = g(x)$ si il existe $n \in \mathbb{N}$ tel que $x = \text{atanh}@@n(-1)$ ou $x = \text{atanh}@@n(1)$ alors $f(x) = x$
On remarquera que f est continue au point $x = 0$

3.3 Cas général

3.3.1 Un lemme

Soit g une injection de E dans $F \subset E$. Alors il existe une bijection f de E dans F.
On note $K_0 = E \setminus F$ et $K_{n+1} = g(K_n) \subset F$ pour $n \in \mathbb{N}$.
Soit $K = \cup K_n$ pour $n \in \mathbb{N}$. On a donc $g(K) \subset K$.
Soit f la fonction de E dans F définit par :
si $x \in K$ alors $f(x) = g(x)$
si $x \notin K$ alors $f(x) = x$

Alors f est une bijection de E dans F.
En effet : f est injective car si $f(x_1) = f(x_2)$ on a 3 cas possibles : $x_1 \in K$ et $x_2 \in K$ alors $f(x_1) = g(x_1) = f(x_2) = g(x_2)$ et g injective entraine $x_1 = x_2$,
$x_1 \notin K$ et $x_2 \notin K$ alors $f(x_1) = x_1 = f(x_2) = x_2$ $f(x_1) = x_1 = f(x_2) = x_2$ donc $x_1 = x_2$,
$x_1 \in K$ et $x_2 \notin K$ alors $f(x_1) = g(x_1) = f(x_2) = x_2$ mais comme $g(K) \subset K$ on en déduit que $x_2 = g(x_1)$ est dans K ce qui est contraire à l'hypothèse.
f est surjective car soit $y \in F \subset E$ on cherche $x \in E$ tel que $f(x) = y$. On a 2 cas :
$y \notin g(K)$ alors $y \notin K$ donc $f(y) = y$,
$y \in g(K)$ alors il existe $x \in K_n$ pour un $n \geq 1$ tel que $y = g(x) = f(x)$ Donc f est une bijection de E dans F.

3.3.2 Le théoreme de Bernstein

Soitent 2 fonctions injectives g_1 de $E1$ dans $E2$ et g_2 de $E2$ dans $E1$.
Alors il existe une bijection h de $E1$ dans $E2$.
Soient $F = g2(E2) \subset E1$ et $g = g2@g1$.
g est une injection de $E1$ dans $F \subset E1$ donc d'après le lemme il existe une bijection f entre $E1$ et F. Or g_2 est une bijection de $E2$ dans F (g_2 est injective et surjective car $F = g2(E2)$).
Soit $k = g_2^{-1}$ l'application de F dans $E2$. k est bijective.
Donc $h = k@f$ est une bijection de $E1$ dans $E2$ comme composé de 2 applications bijectives.

Chapitre 4

La suite de Fibonacci et le nombre d'or

4.1 Des exercices pour commencer

1. De combien de façons peut-on vider un tonneau de n litres avec un pot de 1 litre et un pot de 2 litres ? (2 façons sont identiques si la suite des prélèvements sont identiques par exemple pour $n = 3$ on a (1,1,1), (1,2) et (2,1) soit 3 façons). Soit $u(n)$ le nombre de façons de vider un tonneau de n litres avec un pot de 1 litre et un pot de 2 litres.
   On a :
   $u(0) = 1$ il y a 1 façon de vider un tonneau vide !
   $u(1) = 1$ il y a 1 façon de vider un tonneau de 1 litre : (1).
   $u(2) = 2$ il y a 2 façons de vider un tonneau de 2 litres : (1,1),(2).
   $u(3) = 3$ il y a 3 façons de vider un tonneau de 3 litres : (1,1,1),(2,1),(1,2).
   Soit un tonneau de n litres. Quand on a prélevé 1 litre, il reste à vider un tonneau de $n-1$ litres (cela se termine donc de $u(n-1)$ façons) et quand on a prélevé 2 litres, il reste à vider un tonneau de $n-2$ litres (cela se termine donc de $u(n-2)$ façons)
   Donc :
   $u(n) = u(n-1) + u(n-2)$ avec $u(0) = u(1) = 1$
   On doit donc étudier cette suite récurrente qui s'appelle la suite de Fibonacci
2. On dispose d'un alphabet composé des 2 caractères a et b.
   Avec cet alphabet combien de chaînes de caractères ne comportant pas 2 b à la suite peut-on former ? Soit $v(n)$ ce nombre. On a :
   $v(0) = 1$ il y a 1 façon d'avoir une chaîne vide !
   $v(1) = 2$ on a "a" et "b".
   $v(2) = 3$ on a "aa", "ab" et "ba".
   $v(3) = 5$ on a "aaa", "aab", "aba", "baa" et "bab".
   .
   Soit une chaîne comportant $n \geq 2$ caractères soit elle commence par le caractère a et alors il reste à écrire une chaîne de $n-1$ caractères ne comportant pas 2 b à la suite et soit elle commence par le caractère b ce qui impose qu'elle commence par "ab" et alors il reste à écrire une chaîne de

$n-2$ caractères ne comportant pas 2 b à la suite. Donc :
$v(n) = v(n-1) + v(n-2)$ avec $v(0) = 1$ et $v(1) = 2$
On retrouve la suite de Fibonacci décalé d'un indice : $v(n) = u(n+1)$.
Remarque
On peut remplacer les caractères "a" et "b" par des briques rouges et vertes de Lego et les chaînes de n caractères par une tour de hauteur n briques construite de façon que l'on ne mette pas 2 briques vertes à la suite.
On peut alors demander aux élèves de faire toutes les tours de n briques pour $n = 2, 3, 4$ et de regarder comment on forme toutes les tours de 4 briques à partir des des tours de 3 briques et de 2 briques.
On tape par exemple les programmes pour faire une brique de Lego, un tour de hauteur 3 et une tour de hauteur n :

```
brique(a,c):={
[affichage(carre(a,a+1),c+rempli),carre(a,a+1)]
}:;
tour3(a,c0,c1,c2):={
local L;
L:=NULL;
L:=L,brique(a,c0);
L:=L,brique(a+i,c1);
L:=L,brique(a+2*i,c2);
print(L);
return L;
};
tourn(a,C):={
local L,j,n;
L:=NULL;
n:=size(C);
pour j de 0 jusque n-1 faire
L:=L,brique(a+i*j,C[j]);
fpour;
return L;
};
```

On tape :

```
tour3(-6,1,1,1);
brique(-6,1);
tour3(-4,1,1,2);
tour3(-2,1,2,1);
tour3(0,2,1,1);
tour3(2,2,1,2);
tourn(-6-5*i,[1,1,1,1]);
tourn(-4-5*i,[1,1,2,1]);
tourn(-2-5*i,[1,2,1,1]);
tourn(-5*i,[2,1,1,1]);
tourn(2-5*i,[2,1,2,1]);
tourn(4-5*i,[1,1,1,2]);
tourn(6-5*i,[1,2,1,2]);
```

```
tourn(8-5*i,[2,1,1,2]);
tourn(4,[1,1]);
tourn(6,[1,2]);
tourn(8,[2,1]);
```

On obtient :



Une tour de hauteur 4 se termine :
- soit par une brique rouge,
- soit par une brique verte.

Si on enlève la brique du haut :
- si elle est rouge on obtient une tour de hauteur 3 se terminant soit par une brique rouge, soit par une brique verte.
- si elle est verte, on obtient une tour de hauteur 3 se terminant par une brique rouge et donc en enlevant encore cette brique rouge, on obtient une tour de hauteur 2 se terminant soit par une brique rouge, soit par une brique verte.

On voit sur le dessin qu'une tour de hauteur 4 est composée :
- soit d'une tour de hauteur 3 plus une brique rouge,
- soit d'une tour de hauteur 2 plus une brique rouge et plus une brique verte.

De façon générale si `nbtour(n)` désigne le nombre de tours de hauteur `n`, on a :
`nbtour(n)=nbtour(n-1)+nbtour(n-2)` avec :
`nbtour(1)=2` et `nbtour(2)=3`
Donc `nbtour(3)=2+3=5` et `nbtour(4)=3+5=8`.

Prolongement

Faire un programme `nbtour(n)` qui renvoie le nombre de tours de hauteur `n`.
Faire un programme `couleurtour(n)` qui renvoie une matrice qui a comme ligne les couleurs des tours de hauteur `n`.
Faire un programme `tours(a,n)` qui dessinne les tours de hauteur `n` à partir du point d'affixe `a`.
On tape :

```
brique(a,c):={
[affichage(carre(a,a+1),c+rempli),carre(a,a+1)]
}:;
tourn(a,C):={
local L,j,n;
L:=NULL;
n:=size(C);
pour j de 0 jusque n-1 faire
L:=L,brique(a+i*j,C[j]);
fpour;
return L;
}:;
nbtour(n):={
local a,b,c,k;
si n<0 alors retourne NULL; fsi;
si n==0 alors retourne 1;fsi;
si n==1 alors retourne 2; fsi;
a:=1;
b:=2;
pour k de 2 jusque n faire
    c:=a+b;
    a:=b;
    b:=c;
fpour;
retourne c;
}:;
couleurtour(n):={
local A,L,C,j,k,s0,s1,N;
L:=[[1],[2]],[[1,1],[2,1],[1,2]];
j:=3 ;
N:=nbtour(n);
tantque j<=n faire
   s0:=size(L[0])-1;
   s1:=size(L[1])-1;
   A:=L[0];
   L[0]:=L[1];
   pour k de 0 jusque s1 faire
     L[1,k]:=append(L[1,k],1);
   fpour;
   pour k de 0 jusque s0 faire
     A[k]:=concat(A[k],[1,2]);
   fpour;
   j:=j+1;
   L[1]:=concat(L[1],A);
 ftantque;
 return L[1];
}:;
```

```
tours(a,n):={
 local L,k,R,N;
 R:=NULL;
 L:=couleurtour(n);
 N:=nbtour(n)-1;
 pour k de 0 jusque N faire
 R:=R,tourn(a,L[k]);
 a:=a+2;
 fpour;
 return R;
 }:;
```

On vérifie, on tape :
`nbtour(4),nrows(couleurtour(4)),size(tours(0,4))`
On obtient :
`8,8,32`
32=4*8 est le nombre de briques a dessiner
On ouvre un écran de géométrie et on tape :

```
tours(-3+5i,3);
tours(-6,4);
tours(-12-6i,5);
```

On obtient :



On voit bien comment on a constitué les tours de hauteur 5 :
On rajoute une brique rouge aux tours de hauteur 4 et on rajoute une brique rouge et une brique verte aux tours de hauteur 3.

4.2 La suite de Fibonacci

4.2.1 La définition

La suite de Fibonacci est la suite u définie par :
$u_0 = 0$

$u_1 = 1$
$u_n = u_{n-1} + u_{n-2}$ pour $n > 1$

4.2.2 Le programme avec Xcas

Il ne faut surtout pas écrire un programme récursif car sinon on calcule les mêmes termes plusieurs fois : Pour avoir u_n :

```
Fibonacci(n):={
local a,b,c,k;
si n==0 ou n==1 alors
   retourne n;
fsi;
a:=0;
b:=1;
pour k de 2 jusque n faire
    c:=a+b;
    a:=b;
    b:=c;
fpour;
retourne c;
}:;
```

On tape :
`Fibonacci(10)`
On obtient :
`55`
Pour avoir les n premiers termes (le premier terme est u_0) :

```
Fibonasuite(n):={
local a,b,c,L,k;
L:=NULL;
si n<0 alors retourne L;fsi;
L:=L,0;
si n==0 alors retourne L;fsi;
L:=L,1;
si n==1 alors retourne L; fsi;
a:=0;
b:=1;
pour k de 2 jusque n faire
    c:=a+b;
    L:=L,c;
    a:=b;
    b:=c;
fpour;
retourne L;
}:;
```

On tape :
`Fibonasuite(12)`

On obtient la suite des 11 premiers termes de la suite de Fibonnacci :
`0,1,1,2,3,5,8,13,21,34,55,89,144`

4.2.3 Le programme récursif avec `Xcas`

```
Fibonarec(n):={
si (n==0 or n==1) alors retourne n fsi;
retourne Fibonarec(n-2)+Fibonarec(n-1);
}:;
```

ou encore une écriture récursive avec un `when` :

```
Fibonawhen(n):={
when(n==0 or n==1,n,Fibonawhen(n-2)+Fibonawhen(n-1));
}:;
```

Que faut-il calculer pour calculer `Fibonarec(5)` ?
Il faut calculer `Fibonarec(3)` et `Fibonarec(4)`.
Pour calculer `Fibonarec(3)`, il faut calculer `Fibonarec(1)` et `Fibonarec(2)`.
Pour calculer `Fibonarec(2)`, il faut calculer `Fibonarec(0)` et `Fibonarec(1)`.
Pour calculer `Fibonarec(4)`, il faut calculer `Fibonarec(2)` et `Fibonarec(3)`
donc calculer `Fibonarec(2)` 2 fois et `Fibonarec(3)` 1 fois
Donc pour calculer `Fibonarec(5)` :
`Fibonarec(2)` doit être calculer 3 fois, `Fibonarec(3)` doit être calculer 2 fois et `Fibonarec(4)` doit être calculer 1 fois.
Que faut-il calculer pour calculer `Fibonarec(6)` ?
Il faut calculer `Fibonarec(4)` et `Fibonarec(5)`.
Donc pour calculer `Fibonarec(6)` :
`Fibonarec(5)` doit être calculé 1 fois
`Fibonarec(4)` doit être calculé 1 fois.
`Fibonarec(2)` doit être calculé 2+3=5 fois, `Fibonarec(3)` doit être calculé 2+1=3 fois et `Fibonarec(4)` doit être calculé 2 fois.
On cherche v_n le nombre de fois qu'il faut calculer `Fibonarec(2)` pour calculer `Fibonarec(n)`. On a :
$v_2 = 1$, $v_3 = 1$, $v_4 = 2$, $v_5 = 3$ et puisque pour calculer `Fibonarec(n)`, il faut calculer `Fibonarec(n-2)` et `Fibonarec(n-1)`, on a :
$v_n = v_{n-2} + v_{n-1}$
Donc v_n est la suite de Fibonacci commencant à $n = 2$:
$v_2 = 1, v_3 = 1, v_n = v_{n-2} + v_{n-1}$ i.e. $v_n = u_{n-2}$
On cherche w_n le nombre de fois qu'il faut calculer `Fibonarec(p)` pour calculer `Fibonarec(n)` lorsque p est fixé et vérifie $2 \leq p \leq n$. On a :
$w_p = 1$, $w_{p+1} = 1$ et
puisque pour calculer `Fibonarec(n)`, il faut calculer `Fibonarec(n-2)` et `Fibonarec(n-1)`, on a : $w_n = w_{n-2} + w_{n-1}$
Donc $w_n = v_{n+2-p} = u_{n-p}$

Pour vérifier, on cherche le temps mis pour calculer u_{21}, u_{22}, u_{23} avec la fonction récursive `Fibonarec(n)` `Fibonarec(21)`,`Fibonarec(22)` et `Fibonarec(23)`.
On tape :
`time(Fibonarec(21))`

On obtient
`[1.75,1.802534401]`
Le premier nombre est le temps CPU (temps mis par le processeur pour faire uniquement ces calculs, en seconde), le deuxième nombre est le temps mis pour faire le calcul comme si on chronomètrait.
On tape :
`time(Fibonarec(22))`
On obtient : `[2.79,2.825411609]`
On tape :
`time(Fibonarec(23))`
On obtient : `[4.59,4.614912788]`
On a : `1.75+2.79` renvoie `4.54` qui est proche de `4.59` et `1.802534401+2.825411609` renvoie `4.62794601` qui est proche de `4.614912788`

On modifie `Fibonasuite` en `Fibonarectime` qui donnera le temps mis pour calculer `Fibonarec(0)`...`Fibonarec(n)` en mettant comme 4 premiers termes 0,0,0.00022,0.0003 (car le temps mis pour calculer `Fibonarec(2)` est environ 0.00022 et le temps mis pour calculer et `Fibonarec(3)` est environ 0.0003 :

```
Fibonarectime(n):={
local a,b,c,L,k;
L:=0,0;
si n<=1 alors retourne L;fsi;
L:=L,0.00022;
si n==2 alors retourne L;fsi;
L:=L,0.0003;
si n==3 alors retourne L; fsi;
a:=0.00022;
b:=0.0003;
pour k de 4 jusque n faire
    c:=a+b;
    L:=L,c;
    a:=b;
    b:=c;
fpour;
retourne L;
}:;
```

On tape :
`T:=Fibonarectime(24)`
`T[21],T[22],T[23],T[24]`
On obtient :
`1.82278,2.94932,4.7721,7.72142`
Peut-on espérer calculer `Fibonarec(30)` ? On tape :
`T:=Fibonarectime(30)`
`T[30]`
On obtient :

`138.55526`
cela fait un temps de 138.55526 secondes soit plus de 2 minutes !
Je teste en tapant :
`time(Fibonarec(30))`
On obtient :
`[141.16,142.640424131]`
Le premier nombre est le temps CPU (temps mis par le processeur pour faire uniquement ces calculs, en seconde), le deuxième nombre est le temps mis pour faire le calcul comme si on chronométrait.

4.2.4 Propriétés des termes de la suite de Fibonacci

La suite u_n de Fibonacci définie par :
$u_0 = 0$,$u_1 = 1$ et $u_{n+1} = u_n + u_{n-1}$ pour $n \geq 1$
Voici les 13 premiers termes de cette suite :
$(u_0, u_1, u_2, ..., u_{12})$=(0,1,1,2,3,5,8,13,21,34,55,89,144)
Montrer que cette suite vérifie :

1. $u_0 + u_1 + u_2 + ...u_{n-1} = u_{n+1} - 1$ pour $n \geq 0$
   On fait une démonstration par récurrence :
   On a :
   cette relation est vraie au rang $n = 0$: $u_0 = 0 = 1 - 1 = u_2 - 1$
   si au rang n, on a la relation :
   $u_0 + u_1 + u_2 + ...u_{n-1} = u_{n+1} - 1$ pour $n \geq 0$ alors
   cette relation reste vraie au rang $n + 1$:
   $u_{n+2} - 1 = u_{n+1} + u_n - 1 = u_0 + u_1 + u_2 + ...u_{n-1} + u_n$

2. $u_{n+1}u_{n-1} - u_n^2 = (-1)^n$ pour $n \geq 1$
   En effet :
   pour $n = 1$ on a : $u_2u_0 - u_1^2 = 1 * 0 - 1^2 = -1$,
   pour $n = 2$ on a : $u_3u_1 - u_2^2 = 2 * 1 - 1^2 = 1$,
   Donc si on suppose que la relation R_n :$u_{n+1}u_{n-1} - u_n^2 = (-1)^n$ est vraie, montrons que R_{n+1} est vraie. On a, en remplaçant u_{n+2} par $u_n + u_{n+1}$:
   $u_{n+2}u_n - u_{n+1}^2 = u_nu_n + u_{n+1}u_n - u_{n+1}^2 = u_n^2 + u_{n+1}(u_n - u_{n+1})$
   Donc puisque $u_n - u_{n+1} = -u_{n-1}$
   $u_{n+2}u_n - u_{n+1}^2 = u_n^2 - u_{n+1}u_{n-1} = (-1)^{n+1}$.
   Un paradoxe comme amusement
   Cela conduit au paradoxe suivant :
   Soit un carré de côté 8 unités. On le découpe en 4 morceaux et on dispose ces 4 morceaux comme ci dessous (on pourrait faire la même chose avec un carré de côté un des termes u_n de la suite de Fibonacci puis faire le découpage en utilisant les 2 termes précedents u_{n-2} et u_{n-1}) :



Le carré a comme surface 64 carrés alors que le rectangle est composé de 5*13=65 carrés. D'où vient le carré supplémentaire ? Le carré supplémentaire est l'aire du parallélogramme bleu ! ! ! !



En effet les bords des 4 morceaux ne suivent pas la diagonale du rectangle qui a comme pente $-5/13$ ce qui est différent de -2/5 et de -3/8.
Selon la longueur du carré initial il peut soit y avoir un petit carré en trop soit en manquer 1 puisque $u_{n+1}.u_{n-1} - u_n^2 = (-1)^{n-1}$.
La suite de Fibonnacci vérifie :

$$u_{n+1}.u_{n-1} - u_n^2 = (-1)^{n-1}$$

En effet :
$1 * 2 - 1 = 1, 3 * 1 - 2^2 = -1, 5 * 2 - 3^2 = 1$

et on a :
$u_{n+1}.u_{n-1} - u_n^2 = u_n.u_{n-1} + u_{n-1}^2 - u_n^2 = u_n(u_{n-1} - u_n) + u_{n-1}^2$
Donc :
$u_{n+1}.u_{n-1} - u_n^2 = -u_n.u_{n-2} + u_{n-1}^2$.
On a par exemple :
$5 * 13 - 8^2 = 1$

3. $u_{n+p-1} = u_{n-1}u_{p-1}u_nu_p$ pour $n \geq 1$ et $p \geq 1$
Puisque n et p jouent un rôle symétrique, on va montrer par récurrence sur $n \geq 1$ que pour tout $1 \leq p \leq n$ on a :
$u_{n+p-1} = u_{n-1}u_{p-1} + u_nu_p$
Voici les 13 premiers termes de cette suite :
$(u_0, u_1, u_2, ..., u_{12})$=(0,1,1,2,3,5,8,13,21,34,55,89,144)
Pour $n = 1$ et $p = 1$ on a :
$u_1 = 2 = u_0u_0 + u_1u_1 = 0 + 1 = 1$
donc la relation est vraie.
Pour $n = 2$ et $p = 1$ on a :
$u_2 = u_1u_0 + u_2u_1 = 0 + 1 = 1$
donc la relation est vraie.
Pour $n = 2$ et $p = 2$ on a :
$u_3 = u_1u_1 + u_2u_2 = 1 + 1 = 2$
donc la relation est vraie.
Si la relation R_n est vraie pour n et pour tout p tel que $1 \leq p \leq n$, on a
$u_{n+p-1} = u_{n-1}u_{p-1} + u_nu_p$
Montrons que cette relation est encore vraie pour $n + 1$ à savoir :
pour tout $1 \leq p \geq n + 1$ on a $u_{n+1+p-1} = u_nu_{p-1} + u_{n+1}u_p$
Soit $p \leq n + 1$, on a donc $p - 2 \leq n$ et $p - 1 \leq n$ et
$u_{n+1+p-1} = u_{n+p} = u_{n+p-1} + u_{n+p-2} = u_{n+p-3} + u_{n+p-2} + u_{n+p-2}$
d'après l'hypothèse de récurrence on a puisque $p - 2 \leq n$ et $p - 1 \leq n$:
$u_{n+p-3} = u_{n+(p-2)-1} = u_{n-1}u_{p-3} + u_nu_{p-2}$
$u_{n+p-2} = u_{n+(p-1)-1} = u_{n-1}u_{p-2} + u_nu_{p-1}$
$u_{n+p-3} + 2u_{n+p-2} = u_{n-1}(u_{p-3} + 2u_{p-2}) + u_n(u_{p-2} + 2u_{p-1})$
puisque pour tout n, $u_n + 2u_{n+1} = u_n + u_{n+1} + u_{n+1} = u_{n+1} + u_{n+2}$, on a :
$u_{(n+1)+p-1} = u_{n+p-3} + 2u_{n+p-2} = u_{n-1}(u_{p-2} + u_{p-1}) + u_n(u_{p-1} + u_p)$
$u_{(n+1)+p-1} = u_{n-1}u_p + u_nu_{p+1}$
$u_{n-1} = u_{n+1} - u_n$ donc :
$u_{n+1+p-1} = u_{n+1}u_p - u_nu_p + u_nu_{p+1} = u_{n+1}u_p - u_nu_{p-1}$

4. Montrer que u_n et u_{n+1} sont premiers entre eux.
puisque $u_{n+1} = u_n + u_{n-1}$ un diviseur commun à u_n et u_{n+1} est aussi un diviseur commun à u_n et u_{n-1} ...etc, donc est aussi un diviseur commun à $u_0 = 1$ et $u_1 = 1$.
Donc u_n et u_{n+1} sont premiers entre eux.

5. Déduire des résultats de 3/ et 4/ que pour $m \geq 1$ et $n \geq 1$ on a :
pgcd$(u_m, u_n) = u_p$ avec $p =$pgcd(m, n).
Pour le calcul du pgcd(m, n) on va utiliser l'algorithme soustractif pour calculer le reste d'une division euclidienne (cf le manuel Algorithmes) dont voici les programmes :

```
Iquorem(a,b):={
local q,r;
r:=b;q:=0;
while (a>=b) {
a:=a-b
q:=q+1;
}
return (q,a);
}:;
Pgcd(a,b):={
tantque a!=b faire
  si(a>b) alors
    a:=a-b;
  sinon
    b:=b-a;
  fsi;
ftantque;
retourne a;
}:;
```

Soient 2 entiers $m \geq 1$ et $n \geq 1$ et supposons $m \geq n$.
Posons : $r_1 = m - n \geq 0$.
Si $r_1 = 0$ on a $m = n$ et puisque pgcd$(u_m, u_n) = u_m$, le résultat est vrai.
Si $r_1 \neq 0$ (i.e si $m > n \geq 1$) .
On a d'après 4/ on a puisque $r_1 \geq 1$ et $r_1 + 1 \geq 2$:
$u_m = u_{n+r_1} = u_{n+(r_1+1)-1} = u_{n-1}u_{r_1} + u_n u_{r_1+1}$
si d est un diviseur de u_m et de u_n, d divise $u_{n-1}u_{r_1}$.
Puisque u_n et u_{n-1} sont premiers entre eux, d ne divise pas u_{n-1} donc d divise u_{r_1}.
Si $r_1 < n$ on recommence on avec $r_2 = r_1 - n = m - 2n$,...$r_q = m - nq$ jusqu'à ce que $r_q \leq n$.
On a alors $m = nq + r_q$.
Si $r_q = n$ alors $m = n(q+1)$ et $r_{q+1} = 0$ et pgcd$(u_m, u_n) = u_{r_q} = u_n$ Si $r_q < n$ alors $m = nq + r_q$ donc r_q est le reste de la division euclidienne de m par n (on a aussi q est le quotient de m par n).
Posons $R_1 = r_q$.
On a montré qu'un diviseur de u_m de u_n est aussi un diviseur de u_{R_1} donc pgcd(u_m, u_n) =pgcd(u_n, u_{R_1})
On recommence avec n (jouant le rôle de m) et R_1 (jouant celui de n).
On a $R_1 < n$ et on pose $r_1 = n - R_1$ ....etc, on obtient R_2
Puis on obtient R_3...etc R_{k+1}
On s'arrête lorsque $R_{k+1} = 0$ i.e lorsque R_{k-1} est un multiple de R_k.
On obtient ainsi la suite des restes l'algorithme d'Euclide pour m et n :
$R_1 > R_2 > .. > R_k > R_{k+1} = 0$.
Donc d'après l'algorithme d'Euclide, on a :
p=pgcd$(m, n) = R_k$ et
pgcd(u_m, u_n) =pgcd(u_n, u_{R_1}) =pgcd$(u_{R_1}, u_{R_2}) =$ ..pgcd$(u_{R_{k-1}}, u_{R_k}) = u_{R_k}$.
Donc le pgcd de u_m et u_n est u_p avec $p =$pgcd(m, n).

4.2.5 Exercice

Si F_n est la suite de Fibonnacci définie pour $n > 0$ par :
$u_0 = 0$, $u_1 = 1$, et pour $n \geq 2$ par $u_n = u_{n-1} + u_{n-2}$.
Calculer $s_n = \sum_{k=0}^{n} F_n$ pour $n \in N$
Puisque les 10 premiers termes de u_n sont :
`0,1, 1, 2, 3, 5, 8,13,21,34,55`
les 10 premiers termes de s_n sont :
`0,1, 2, 4, 7,12,20,32,33,54,88`
Montrons par récurrence que pour $n \in \mathbb{N}$ on a $s_n = F_{n+2} - 1$.
On a :
$s_0 = F_2 - 1 = 1 - 1 = 0$ Si $s_n = F_{n+2} - 1$ on a :
$s_{n+1} = s_n + F_{n+1} = F_{n+2} - 1 + F_{n+1} = F_{n+3} - 1$ Donc pour $n \in N$ on a :

$$s_n = \sum k = 0^n F_n = F_{n+2} - 1$$

4.3 La suite de Fibonacci et le triangle de Pascal

4.3.1 Le triangle de Pascal

Pour avoir les premières valeurs du triangle de Pascal, on tape :
`A:=makemat((j,k)->comb(j,k),11,11)`
On obtient :

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 1 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 1 & 2 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 1 & 3 & 3 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 1 & 4 & 6 & 4 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 1 & 5 & 10 & 10 & 5 & 1 & 0 & 0 & 0 & 0 & 0 \\ 1 & 6 & 15 & 20 & 15 & 6 & 1 & 0 & 0 & 0 & 0 \\ 1 & 7 & 21 & 35 & 35 & 21 & 7 & 1 & 0 & 0 & 0 \\ 1 & 8 & 28 & 56 & 70 & 56 & 28 & 8 & 1 & 0 & 0 \\ 1 & 9 & 36 & 84 & 126 & 126 & 84 & 36 & 9 & 1 & 0 \\ 1 & 10 & 45 & 120 & 210 & 252 & 210 & 120 & 45 & 10 & 1 \end{pmatrix}$$

On peut aussi utiliser le tableur et la relation :
pour $n \in \mathbb{N}$: $C_n^0 = 1$, $C_n^n = 1$ et $C_n^p = C_{n-1}^p + C_{n-1}^{p-1}$ pour $0 < p < n$.
On ouvre un niveau tableur et on choisit 22 lignes et 11 colonnes.
On met `1` dans $A0$.
Puis, on copie $A0$ vers le bas pour avoir des `1` dans la colonne A.
Dans $B1$ on met `=B0+A0` formule que l'on copie vers le bas et vers la droite.
On copie alors vers le bas la formule située dans $C1$,...$K1$.
Remarque On peut mettre `idn(11)` dans $A0$, et si on a coché sur `Distribuer`, cela a pour effet de remplir 11 lignes et 11 colonnes avec la matrice identité d'ordre 11. Mais cela est inutile car les 1 de la diagonale sont recalculés quand on copie vers la droite !
On obtient :

$$\begin{pmatrix}
1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
1 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
1 & 2 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
1 & 3 & 3 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
1 & 4 & 6 & 4 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\
1 & 5 & 10 & 10 & 5 & 1 & 0 & 0 & 0 & 0 & 0 \\
1 & 6 & 15 & 20 & 15 & 6 & 1 & 0 & 0 & 0 & 0 \\
1 & 7 & 21 & 35 & 35 & 21 & 7 & 1 & 0 & 0 & 0 \\
1 & 8 & 28 & 56 & 70 & 56 & 28 & 8 & 1 & 0 & 0 \\
1 & 9 & 36 & 84 & 126 & 126 & 84 & 36 & 9 & 1 & 0 \\
1 & 10 & 45 & 120 & 210 & 252 & 210 & 120 & 45 & 10 & 1 \\
1 & 11 & 55 & 165 & 330 & 462 & 462 & 330 & 165 & 55 & 11 \\
1 & 12 & 66 & 220 & 495 & 792 & 924 & 792 & 495 & 220 & 66 \\
1 & 13 & 78 & 286 & 715 & 1287 & 1716 & 1716 & 1287 & 715 & 286 \\
1 & 14 & 91 & 364 & 1001 & 2002 & 3003 & 3432 & 3003 & 2002 & 1001 \\
1 & 15 & 105 & 455 & 1365 & 3003 & 5005 & 6435 & 6435 & 5005 & 3003 \\
1 & 16 & 120 & 560 & 1820 & 4368 & 8008 & 11440 & 12870 & 11440 & 8008 \\
1 & 17 & 136 & 680 & 2380 & 6188 & 12376 & 19448 & 24310 & 24310 & 19448 \\
1 & 18 & 153 & 816 & 3060 & 8568 & 18564 & 31824 & 43758 & 48620 & 43758 \\
1 & 19 & 171 & 969 & 3876 & 11628 & 27132 & 50388 & 75582 & 92378 & 92378 \\
1 & 20 & 190 & 1140 & 4845 & 15504 & 38760 & 77520 & 125970 & 167960 & 184756 \\
1 & 21 & 210 & 1330 & 5985 & 20349 & 54264 & 116280 & 203490 & 293930 & 352716
\end{pmatrix}$$

On a ainsi les coefficients binomiaux utiles pour le développement de $(1+x)^n$ pour $n = 0..21$.

4.3.2 La somme des diagonales montantes

Lorsqu'on fait la somme des diagonales montantes du triangle de Pascal on obtient la suite de Fibonacci.

Par exemple :

$$\begin{pmatrix}
1 & = \\
 & \\
1 & = \\
 & \\
2 & = \\
 & \\
3 & = \\
 & \\
5 & = \\
 & \\
8 & = \\
 & \\
13 & =
\end{pmatrix}
\begin{pmatrix}
1 & & 0 & & 0 & & 0 & & 0 & & 0 \\
 & \nearrow & & \nearrow & & \nearrow & & \nearrow & & \nearrow & \\
1 & & 1 & & 0 & & 0 & & 0 & & 0 \\
 & \nearrow & & \nearrow & & \nearrow & & \nearrow & & \nearrow & \\
1 & & 2 & & 1 & & 0 & & 0 & & 0 \\
 & \nearrow & & \nearrow & & \nearrow & & \nearrow & & & \\
1 & & 3 & & 3 & & 1 & & 0 & & 0 \\
 & \nearrow & & \nearrow & & \nearrow & & & & & \\
1 & & 4 & & 6 & & 4 & & 1 & & 0 \\
 & \nearrow & & \nearrow & & & & & & & \\
1 & & 5 & & 10 & & 10 & & 5 & & 1 \\
 & \nearrow & & & & & & & & & \\
1 & & 6 & & 15 & & 20 & & 15 & & 6
\end{pmatrix}$$

On tape :

```
L:=sum(A[j-k, k],k=0..j)$(j=0..10)
```

On obtient :

```
1,1,2,3,5,8,13,21,34,55,89
```

Pour le montrer on utilise les relations pour $n \in \mathbb{N}$ et $p \in \mathbb{N}$:

$C_n^0 = 1$, $C_n^n = 1$, $C_n^p = 0$ si $p > n$ et $C_n^p = C_{n-1}^p + C_{n-1}^{p-1}$ pour $0 < p \leq n$.
On tape :
`simplify(comb(j-1,k-1)+comb(j-1,k)-comb(j,k))`
On obtient :
`0`
Soit a_n la suite définie par la somme des diagonales montantes du triangle de Pascal.
On a donc :
$a_0 = 1$, $a_1 = 1$ et pour $n > 1$:
$a_n = \sum_{p=0}^{\texttt{floor(n/2)}} \texttt{comb}(\texttt{n}-\texttt{p},\texttt{p}) = 1 + \sum_{p=1}^{\texttt{floor(n/2)}} \texttt{comb}(\texttt{n}-\texttt{p},\texttt{p})$
$a_n = 1 + \sum_{p=1}^{\texttt{floor(n/2)}} \texttt{comb}(\texttt{n}-\texttt{p}-\texttt{1},\texttt{p}-\texttt{1}) + \sum_{p=1}^{\texttt{floor(n/2)}} \texttt{comb}(\texttt{n}-\texttt{p}-\texttt{1},\texttt{p})$
On a pour tout n : $\texttt{floor}(\texttt{n}/2) - \texttt{1} = \texttt{floor}((\texttt{n}-2)/2))$.
On a si $n = 2k$, on a $\texttt{floor}(\texttt{n}/2) = \texttt{floor}((\texttt{n}-\texttt{1})/2) + \texttt{1} = k$:
$1 + \sum_{p=1}^{\texttt{floor(n/2)}} \texttt{comb}(\texttt{n}-\texttt{p}-\texttt{1},\texttt{p}) =$
$\sum_{p=0}^{\texttt{floor((n-1)/2)}} \texttt{comb}(\texttt{n}-\texttt{1}-\texttt{p},\texttt{p}) + \texttt{comb}(\texttt{2k}-\texttt{1}-\texttt{k},\texttt{k}) = a_{n-1}$
car $\texttt{comb}(\texttt{2k}-\texttt{1}-\texttt{k},\texttt{k}) = 0$
$\sum_{p=1}^{\texttt{floor(n/2)}} \texttt{comb}(\texttt{n}-\texttt{p}-\texttt{1},\texttt{p}-\texttt{1}) =$
$\sum_{p=0}^{\texttt{floor((n-2)/2)}} \texttt{comb}(\texttt{n}-\texttt{2}-\texttt{p},\texttt{p}) = a_{n-2}$
car `floor(n/2)-1=floor((n-2)/2)`
On a si $n = 2k + 1$ alors $\texttt{floor}(\texttt{n}/2) = \texttt{floor}((\texttt{n}-\texttt{1})/2) = \texttt{k}$:
$1 + \sum_{p=1}^{\texttt{floor(n/2)}} \texttt{comb}(\texttt{n}-\texttt{p}-\texttt{1},\texttt{p}) =$
$\sum_{p=0}^{\texttt{floor((n-1)/2)}} \texttt{comb}(\texttt{n}-\texttt{1}-\texttt{p},\texttt{p}) = a_{n-1}$
$\sum_{p=1}^{\texttt{floor(n/2)}} \texttt{comb}(\texttt{n}-\texttt{p}-\texttt{1},\texttt{p}-\texttt{1}) =$
$\sum_{p=0}^{\texttt{floor((n-2)/2)}} \texttt{comb}(\texttt{n}-\texttt{2}-\texttt{p},\texttt{p}) = a_{n-2}$
car `floor(n/2)-1=floor((n-2)/2)`
Donc :
$a_0 = 1$, $a_1 = 1$ et pour $n > 1$ on a $a_n = a_{n-1} + a_{n-2}$
a_n est donc une suite de Fibonacci définie par :
$u_0 = u_1 = 1$,et pour $n \geq 2$ $u_n = u_{n-2} + u_{n-1}$.
On tape :
`L1:=0,sum(A[j-k-1,k-1],k=1..j-1)$(j=1..10)`
On obtient :
`0,0,1,1,2,3,5,8,13,21,34`
On tape :
`L2:=1,sum(A[j-k-1,k],k=0..j-1)$(j=1..10)`
On obtient :
`1,1,1,2,3,5,8,13,21,34,55`
On tape :
`[L1]+[L2]`
On obtient :
`[1,1,2,3,5,8,13,21,34,55,89]`
On tape :
`simplify(comb(j-1,k-1)+comb(j-1,k)-comb(j,k))`
On obtient :
`0`

4.3.3 Si F_n est la suite de Fibonnacci, calcul de $\sum_{k=0}^{n} C_n^k F_k$

Soit F_n la suite de Fibonnacci : 1,1,2,3,5,8,13,21,13+21=34... définie par : $u_0 = u_1 = 1$,et pour $n \geq 2$ $u_n = u_{n-2} + u_{n-1}$.

On veut calculer $s_n = \sum_{k=0}^{n} C_n^k F_k$.
$s_0 = 1$
$s_1 = 1 + 1 = 2 = F_2$
$s_2 = 1 + 2 * 1 + 1 * 2 = 5 = F_4$
$s_3 = 1 + 3 * 1 + 3 * 2 + 1 * 3 = 13 = F_6$
Il semble donc que $s_n = \sum_{k=0}^{n} C_n^k F_k = F_{2n}$.
On va montrer cette formule par récurrence, mais pour cela il faut aussi connaitre la somme $\sigma_n = \sum_{k=0}^{n} C_n^k F_{k+1}$.
On a :
$\sigma_0 = 1$.
$\sigma_1 = 1 + 2 = 3 = F_3$.
$\sigma_2 = 1 + 2 * 2 + 1 * 3 = 8 = F_5$.
$\sigma_3 = 1 + 3 * 2 + 3 * 3 + 1 * 5 = 21 = F_7$.
Il semble donc que $\sigma_n = \sum_{k=0}^{n} C_n^k F_{k+1} = F_{2n+1}$.
Montrons ces deux formules par récurrence et supposons que pour tout $p \leq n$ on a : $s_p = \sum_{k=0}^{p} C_p^k F_k = F_{2p}$ et $\sigma_p = \sum_{k=0}^{p} C_p^k F_{k+1} = F_{2p+1}$
Ces deux formules sont-elles vraies pour $p = n + 1$?
On a :
$s_{n+1} = \sum_{k=0}^{n} C_{n+1}^k F_k + F_{n+1}$ et
$\sigma_{n+1} = \sum_{k=0}^{n} C_{n+1}^k F_{k+1} + F_{n+2}$
On sait que :
$C_{n+1}^0 = C_n^0 = C_{n+1}^{n+1} = C_n^n = 1$ et
pour $0 < k < n + 1$ $C_{n+1}^k = C_n^k + C_n^{k-1}$ donc
$s_{n+1} = \sum_{k=0}^{n} C_n^k F_k + \sum_{k=1}^{n} C_n^{k-1} F_k + F_{n+1}$.
D'après les hypothèses de récurrence, on a :
$\sum_{k=0}^{n} C_n^k F_k = s_n = F_{2n}$ et
$\sum_{k=1}^{n} C_n^{k-1} F_k + F_{n+1} = \sum_{j=0}^{n-1} C_n^j F_{j+1} + F_{n+1} = \sum_{j=0}^{n} C_n^j F_{j+1} = \sigma_n = F_{2n+1}$
Donc on a :
$s_{n+1} = F_{2n} + F_{2n+1} = F_{2n+2}$
La formule est valable pour s_{n+1}.
Pour σ_{n+1}, on a :
$\sigma_{n+1} = \sum_{k=0}^{n} C_n^k F_{k+1} + \sum_{k=1}^{n} C_n^{k-1} F_{k+1} + F_{n+2}$
D'après les hypothèses de récurrence, on a :
$\sum_{k=0}^{n} C_n^k F_{k+1} = \sigma_n$ et
$\sum_{k=1}^{n} C_n^{k-1} F_{k+1} + F_{n+2} = \sum_{j=0}^{n-1} C_n^j F_{j+2} + F_{n+2} = \sum_{j=0}^{n} C_n^j F_{j+2} = \sum_{j=0}^{n} C_n^j F_j + \sum_{j=0}^{n} C_n^j F_{j+1} = s_n + \sigma_n = F_{2n} + F_{2n+1} = F_{2n+2}$
Donc :
$\sigma_{n+1} = \sigma_n + F_{2n+2} = F_{2n+1} + F_{2n+2} = F_{2n+3}$
La formule est valable pour σ_{n+1}.
Donc on a pour tout n :

$$s_n = \sum_{k=0}^{n} C_n^k F_k = F_{2n}$$

$\sigma_n = \sum_{k=0}^{n} C_n^k F_{k+1} = F_{2n+1}$

4.4 Le code de Fibonacci

À partir de "sa" suite, Fibonacci a fabriqué un code qui permet d'écrire tous les entiers avec des 0 et des 1. Pour cela, si u_n est la suite de Fibonacci $u_0 = u_1 = 1$ et pour $n \geq 2$ $u_n = u_{n-2} + u_{n-1}$, le code de 0 est 0 et pour $n > 0$ le code des nombres u_n est 1 suivi de $n-1$ zéros.
Si $u_n < m < u_{n+1}$ alors $m = u_n + p$ et comme on a $0 < p = m - u_n < u_{n-1}$ le code de m la suite de Fibonacci est : 1,1,2,3,5,8,13,21,34... On a ainsi :
0 a pour code 0
1 a pour code 1
2 a pour code 10
3 a pour code 100
4 a pour code 101 car 4=3+1
5 a pour code 1000
6 a pour code 1001 car 6=5+1
7 a pour code 1010 car 7=5+2
8 a pour code 10000
etc...
12 a pour code 10101 car 12=8+4=8+3+1
etc...
On remarquera que dans ce code il n'y a jamais deux 1 qui se suivent, en effet :
$u_{n-1} + u_{n-2} = u_n$.
Par exemple le nombre :
$u_1 + u_2 + u_3 + u_4 = 1 + 2 + 3 + 5 = 11$ a comme code 1010 car 11=8+3.
Exercice
Écrire un programme qui renvoie le code de Fibonacci d'un entier $n \geq 10^8$.
On remarquera que $u_{38} < 10^8 < u_{39}$.
On construit la liste des 40 premiers termes de u_n pour $n = 0, 1..39$.
Le codage
On utilise le programme vu précédemment :

```
Fibonasuite(n):={
local a,b,c,L;
L:=NULL;
si n<=0 alors retourne L;fsi;
L:=L,1;
si n==1 alors retourne L;fsi;
L:=L,1;
si n==2 alors retourne L; fsi;
a:=1;
b:=1;
pour k de 3 jusque n faire
    c:=a+b;
    L:=L,c;
    a:=b;
```

```
    b:=c;
fpour;
retourne L;
}:;
```

On écrit le code d'un nombre n au moyen d'une chaîne de caractères : c'est `string` qui transforme l'écriture d'un nombre en une chaîne de caractères

```
codefib(n):={
local fib,cod,p;
si n>10^8 alors retourne "l'entier doit etre <10^8" fsi;
si (n==0 or n==1) alors  return n fsi;
fib:=Fibonasuite(40);
cod:=0;
repeter
p:=2;
tantque n>=fib[p] faire p:=p+1 ftantque;
// fib[p-1]<=n<fib[p]
cod:=10^(p-2)+cod;
n:=n-fib[p-1];
jusqua n==0;
retourne string(cod);
}:;
```

On tape :
`codefib(n)$(n=0..20)`
On obtient une suite de chaînes :
`0,1,10,100,101,1000,1001,1010,10000,10001,10010,10100,10101,`
`100000,100001,100010,100100,100101,101000,101001,101010`
Le décodage
On donne une chaîne de caractères L de 1 et de zéro qui est le code de Fibonacci d'un nombre n et on obtient le nombre n codé par L.
On remarquera que le décodage ci-dessous ne vérifie pas que L est bien un code de Fobonacci : il peut y avoir 2 "1" à la suite (par ex decodefib("11") renverra 3) et les chiffres autres que "1" seront considérés comme des "0" (par ex decodefib("15") renverra 2).
On tape :

```
decodefib(L):={
local k,n,s,fib;
fib:=Fibonasuite(40);
s:=dim(L);
si s==1 alors return expr(L[0]); fsi;
n:=0;
pour k de 0 jusque s-1 faire
si L[k]=="1" alors n:=n+fib[s-k];fsi;
fpour;
return n;
}:;
```

On tape :
`decodefib("101010"),decodefib("10101")`
On obtient :
`20,12`

4.5 Le crible de Fibonacci

Fibonnacci a effectué le crible suivant :
Il écrit la suite des nombres de 1 à n.
Puis il parcourt cette suite et il barre les nombres qui sont les doubles des nombres qui n'ont pas été barrés.
Par exemple si $n = 10$:
il barre 2 (2 est le double de 1) et il ne barre pas 4 car 2 a été barré,
il barre 6, 8 et 10 car 3, 4, 5 n'ont pas été barrés et il obtient :
(1,3,5,7,9).
Écrire un programme qui effectue ce crible.
Caractériser les nombres de ce crible et utiliser cette caractéristion pour écrire une fonction booléenne que teste si un entier est dans ce crible.
Utiliser cette fonction booléenne pour écrire un programme qui effectue ce crible.

Les nombres impairs sont dans le crible.
Un nombre pair $2 * p$ est dans le crible si p n'est pas dans le crible.
On tape :

```
fibocrible(n):={
  local k,L;
  L:=NULL;
   pour k de 1 jusque n faire
    si est_impair(k) alors
      L:=L,k;
    sinon
      si member(iquo(k,2),[L])==0 alors L:=L,k; fsi;
    fsi;
  fpour;
  retourne L;
}:;
```

On tape :
`fibocrible(50)`
On obtient :
`1,3,4,5,7,9,11,12,13,15,16,17,19,20,21,23,25,27,28,29,31,33,35,36,37,39,`
`41,43,44,45,47,48,49`
Les nombres du crible sont les nombres impairs et les nombres pair de la forme
$4^j * (2 * k + 1)$ avec $(j, k) \in \mathbb{N}$ On tape :

```
estdscriblef(p):={
     si est_impair(p) alors retourne vrai; fsi;
    tantque irem(p,4)==0 faire p:=iquo(p,4); ftantque;
    si est_impair(p) alors retourne vrai; sinon retourne faux fsi;
```

```
    }:;
```

On tape :
`estdscriblef(256)`
On obtient :
`vrai`
On tape :
`estdscriblef(250)`
On obtient :
`faux`
On tape le programme qui utilse `estdscriblef` :

```
fiboestcrible(n):={
  local k,L;
  L:=NULL;
   pour k de 1 jusque n faire
    si est_impair(k) alors
      L:=L,k;
    sinon
      si estdscriblef(k) alors L:=L,k; fsi;
    fsi;
  fpour;
  retourne L;
}:;
```

On tape :
`fiboestcrible(50)`
On obtient :
`1,3,4,5,7,9,11,12,13,15,16,17,19,20,21,23,25,27,28,29,31,33,35,36,37,`
`41,43,44,45,47,48,49`

4.6 Le nombre d'or

Les quotients des termes successifs de la suite de Fibonnacci est la suite $v_n = \frac{u_{n+1}}{u_n}$ pour $n \geq 0$
Cette suite converge vers $\frac{1+\sqrt{5}}{2}$: c'est le nombre d'or qui est noté Φ.
En effet, on a $v_0 = 1$ et $v_n = 1 + \frac{u_{n-1}}{u_n} = 1 + \frac{1}{v_{n-1}}$ donc
$v_1 = 2$, $v_2 = 3/2$, $v_3 = 5/3$.
Montrons par reécurrence que $1 \leq v_n \leq 2$ pour tout n : $1 \leq v_0 = 1 < 2$ si $1 \leq v_{n-1} < 2$ alors $1/2 \leq \frac{1}{v_{n-1}} \leq 1$ donc $1 \leq 1 + 1/2 \leq v_n \leq 1 + 1 = 2$
Étude du signe de $v_n - v_{n-1}$:
$v_n - v_{n-1} = \frac{1}{v_{n-1}} - \frac{1}{v_{n-2}} = \frac{v_{n-2}-v_{n-1}}{v_{n-1}v_{n-2}}$
Donc $v_n - v_{n-1}$ est du signe opposé à celui de $v_{n-1} - v_{n-2}$.
Donc $v_{2n} - v_{2n-2}$ a le mm̂e signe que $v_2 - v_0 = 1/2 > 0$ et
$v_{2n+1} - v_{2n-1}$ a le même signe que $v_3 - v_1 = -1/3 < 0$
La suite v_{2n} est convergente vers $1 \leq a \leq 2$ car croissante et majorée.
v_{2n+1} est convergente vers $1 \leq b \leq 2$ car décroissante et minorée.
On doit avoir $a = 1 + 1/b$ et $b = 1 + 1/a$ donc $ab = b + 1 = a + 1$ donc $a = b$

Donc v converge vers $a = b = \phi$ qui vérifie :
$\phi = 1 + \frac{1}{\phi}$ ou encore $\phi^2 - \phi - 1 = 0$ et $1 \leq \phi \leq 2$. donc $\phi = \frac{1+\sqrt{5}}{2}$ Le nombre d'or est :

$$\phi = \frac{1+\sqrt{5}}{2} \simeq 1.61803398875$$

4.6.1 Propriétés du nombre d'or

$$\phi^2 = 1 + \phi$$

$$(\phi - 1) * \phi = 1$$

$$\phi = \sqrt{1 + \sqrt{1 + \sqrt{1 + \sqrt{1 + ...}}}}$$

$$\frac{1}{\phi} = 1 + \phi$$

$$\phi = 1 + \cfrac{1}{1 + \cfrac{1}{1 + \cfrac{1}{1 + ...}}}$$

Si un réctangle est tel que le rapport de la longueur L à la largeur l soit ϕ alors :

$$\frac{L}{l} = \frac{l}{L-l} = \phi$$

On tape :

```
spiror(a,b,n):={
local au,r,L;
L:=NULL;
si n==0 alors retourne L,segment(a,b);fsi;
au :=(1+sqrt(5))/2;
L:=L,rectangle(a,b,au-1);
r:=(b-a)*(au-1);
L:=L,affichage(cercle(a+r,r,pi/2,pi),1);
retourne L,spiror(a+r*(1+i),a+r,n-1);
}:;
```

puis on tape :
`spiror(0,1/2+sqrt(5)/2,7)`
On obtient :



4.7 Un exercice niveau troisième

On veut construire un rectangle d'or i.e. un rectangle tel que :
longueur/largeur $=\frac{1+\sqrt{5}}{2} = \Phi$.
Tout d'abord étant donné un segment AB, on construit un point M tel que $\frac{AB}{MB} = \frac{MB}{MA} = \frac{1+\sqrt{5}}{2} = \Phi$
Soient a un nombre réel positif et ABC un triangle rectangle en A tel que $AB = a$ et $AC = a/2$.
Soit N intersection du cercle de centre C et de rayon AC.
Soit M intersection du cercle de centre B et de rayon BN.



Calculer les longueurs BC, BN, BM et AM.
Montrer que M partage BA selon le nombre d'or i.e. :

$$\frac{MB}{MA} = \frac{AB}{MB} = \frac{1+\sqrt{5}}{2} = \Phi$$

En déduire une construction d'un rectangle d'or avec `Xcas`
Solution
D'après le théorème de Phytagore on a :
$BC^2 = AB^2 + AC^2 = a^2 + a^2/4 = 5a^2/4$
Donc :
$BC = \dfrac{a\sqrt{5}}{2}$
$BN = BM = BC - CN = BC - a/2 = \dfrac{a(\sqrt{5}-1)}{2}$
$AM = AB - BM = a - \dfrac{a(\sqrt{5}-1)}{2} = \dfrac{a(3-\sqrt{5})}{2}$
On a donc :

$$\frac{MB}{MA} = \frac{\sqrt{5}-1}{3-\sqrt{5}} = \frac{1+\sqrt{5}}{2} = \Phi$$

$$\frac{AB}{MB} = \frac{2}{\sqrt{5}-1} = \frac{1+\sqrt{5}}{2} = \Phi$$

Remarque
On peut se servir de `Xcas` pour faire les calculs.
On peut stocker les valeurs dans des variables. On tape :
`a^2+a^2/4`
On obtient BC^2 :
`5*a^2/4`
On tape :
`factor(sqrt(5*a^2/4))`
On obtient BC :
`sqrt(5)*a/2`
On tape :
`factor(a*sqrt(5)/2-a/2)`
On obtient CN et BM :
`(sqrt(5)-1)*a/2`
On tape :
`factor(a-(sqrt(5)-1)*a/2)`
On obtient AM :
`(-(sqrt(5))+3)*a/2`
On tape :
`normal((sqrt(5)-1)*a/2/((-(sqrt(5))+3)*a/2))`
On obtient BM/AM :
`(sqrt(5)+1)/2`
On tape :
`normal(a/((sqrt(5)-1)*a/2))`
On obtient AB/BM :
`(sqrt(5)+1)/2`.
On peut stocker les valeurs dans des variables pour que les calculs soient plus lisibles.
On tape :
`BC:=factor(sqrt(a^2+a^2/4))`
On obtient BC :
`sqrt(5)*a/2`

On tape :
`BM:=factor(BC-a/2)`
On obtient CN et BM :
`a*(sqrt(5)-1)/2`
On tape :
`AM:=factor(a-BM)`
On obtient AM :
`a*(-(sqrt(5))+3)/2`
On tape :
`normal(BM/AM)`
On obtient BM/AM :
`(sqrt(5)+1)/2`
On tape :
`normal(a/BM)`
On obtient AB/BM :
`(sqrt(5)+1)/2.`

Construction du rectangle d'or

```
A:=point(0);
B:=point(4);
K:=point(0,2,affichage=3);
triangle(A,B,K,affichage=3);
cercle(K,2,affichage=3);
N:=inter(cercle(K,2),segment(K,B));
l:=longueur(K,B);
cercle(B,l-2,affichage=3);
M:=inter(cercle(B,l-2),segment(A,B));
rectangle(A,B,(l-2)/4,D,C);
L:=point(6-l,l-2);
segment(M,L);
```

On obtient :



Les rectangles $ABCD$ et $AMLD$ sont des rectangles d'or et $MBCL$ est un carré.

4.8 Un exercice niveau terminale

4.8.1 L'énoncé

On cosidère la fonction f definie sur $\mathbb{R}^*$ par :

$$f(x) = \frac{x+1}{x}$$

On note $f@@n = f \circ f .. \circ f$ (On compose f, n fois)

1. Calculer : $f@@2 = f \circ f, \quad f@@3 = f \circ f \circ f, \quad f@@4 = f \circ f \circ f \circ f$
2. Trouver la valeur de $f@@n$ en fonction de n.
3. Résoudre, lorsqu'on utilise n traits de fractions, l'équation d'inconnue x :

$$x = 1 + \frac{1}{1 + \frac{1}{1+\frac{1}{1+\frac{1}{..+\frac{1}{1+\frac{1}{x}}}}}}$$

4. Touver la limite lorsque n tends vers $+\infty$ de

$$1 + \frac{1}{1 + \frac{1}{1+\frac{1}{1+\frac{1}{...+\frac{1}{1+\frac{1}{a}}}}}}$$

lorsque $a > 0$ et lorsqu'on utilise n traits de fractions.

4.8.2 La solution avec `Xcas`

1. On définit la fonction f, on tape :
   `f(x):=(x+1)/x`
   On calcule $f@@2 = f(f(x))$, on tape :
   `normal((f@@2)(x))`
   On obtient :
   `(2*x+1)/(x+1)`
   On calcule $f@@3 = f(f(f(x)))$, on tape :
   `normal((f@@3)(x))`
   On obtient :
   `(3*x+2)/(2*x+1)`
   On calcule $f@@4$, on tape :
   `normal((f@@4)(x))`
   On obtient :
   `(5*x+3)/(3*x+2)`

2. On suppose que : $f@@n = \dfrac{a_n x + b_n}{c_n x + d_n}$
   On cherche une relation de récurrence entre les différents coefficients.
   On définit la fonction g, on tape :
   `g(x,a,b,c,d):=(a*x+b)/(c*x+d)`
   On a : $(f@@n+1)(x) = f(g(x,a,b,c,d))$
   On calcule $f(g(x,a,b,c,d))$ et on tape :
   `normal(f(g(x,a,b,c,d)))`
   On obtient :
   `(a*x+b+c*x+d)/(a*x+b)`
   donc :
   $c_{n+1} = a_n, \quad d_{n+1} = b_n$
   $a_{n+1} = a_n + c_n = a_n + a_{n-1}$
   $b_{n+1} = b_n + d_n = b_n + b_{n-1}$
   On sait que $a_1 = 1,\ b_1 = 1\ a_0 = c_1 = 1\ b_0 = d_1 = 0$
   Donc si la suite de Fibonacci est la suite u définie par :
   $u_0 = 1, \quad u_1 = 1$
   $u_n = u_{n-1} + u_{n-2}$ pour $n > 1$:
   Alors $f@@n = \dfrac{a_n x + b_n}{c_n x + d_n}$ avec $a_n = u_n,\ b_n = c_n = a_{n-1},\ d_n = a_{n-2}$

3. On résout l'équation d'inconnue x, avec 1 trait de fractions : $x = 1 + \dfrac{1}{x}$
   On tape :
   `normal(1+1/x)`
   On obtient : `(x+1)/x`
   Donc $f(x) = 1 + \dfrac{1}{x}$
   Il faut donc résoudre $f(x) = x$, on tape :
   `solve(x=f(x),x)`
   On obtient : `[1/2*(1-sqrt(5)),1/2*(1+sqrt(5))]`
   On reconnait le nombre d'or et l'inverse de son opposé.
   On résout l'équation d'inconnue x, avec 2 traits de fractions : $x = 1 +$

$\dfrac{1}{1+\dfrac{1}{x}}$
On tape :
`normal(1+1/(1+1/x))`
On obtient : `(2*x+1)/(x+1)`
On reconnait $f(f(x))$ ou bien puisque $f(x) = 1 + \dfrac{1}{x}$, on a :
$1 + \dfrac{1}{1+\dfrac{1}{x}} = 1 + \dfrac{1}{f(x)} = f(f(x))$
. Il faut donc résoudre $f(f(x)) = x$, on tape :
`solve(f(f(x))=x,x)`
On obtient : `[1/2*(1-sqrt(5)),1/2*(1+sqrt(5))]`
qui sont les mêmes solutions qur $f(x) = x$.
On doit résoudre l'équation avec n traits de fractions : $(f@@n)(x) = x$
Comme $(f@@n)$ est une fonction homographique (i.e. $(f@@n)(x)$ est de la forme $(a*x+b)/(c*x+d)$)), cette équation est une équation du 2-nd degré donc admet au plus 2 solutions. Les 2 solutions de $f(x) = x$ sont aussi solutions de $(f@@n)(x) = x$ donc $(f@@n)(x) = x$ a les mêmes solutions que $f(x) = x$.

4. Chercher la limite lorsque le nombre de traits tend vers l'infini de :

$$1 + \cfrac{1}{1 + \cfrac{1}{1 + \cfrac{1}{1 + \cfrac{1}{\cdots \frac{1}{1+\frac{1}{a}}}}}}$$

revient à chercher la limite de la suite des itérées de f définit par : $u_0 = a > 0,\ u_n = f(u_{n-1})$ pour $n > 0$. On a en effet (avec n traits de fractions) :

$$u_n = 1 + \cfrac{1}{1 + \cfrac{1}{1 + \cfrac{1}{1 + \cfrac{1}{\cdots \frac{1}{1+\frac{1}{a}}}}}}$$

. Avec `Xcas`, on tape dans un niveau de géométrie 2-d :
`supposons(a=[6.0,0,9,0.1])`
`plotseq(1+1/x,[a,0,9],5)`
On obtient :



Montrons que la suite u_n converge vers $\phi = \frac{1+\sqrt{5}}{2}$ qui est la solution positive de l'équation $f(x) = x$.
Puisque $a > 0$, on a $u_1 = b = 1 + 1/a > 1 > 0$ et pour tout $n > 0$ $u_n > 1 > 0$.
Si $u_0 = a > \phi$ alors $b = u_1 = 1 + 1/a < 1 + 1/\phi = \phi$ et
si $u_0 = a < \phi$ alors $b = u_1 = 1 + 1/a > 1 + 1/\phi = \phi$ De même si $c = u_n > \phi$ alors $d = u_{n+1} = 1 + 1/c < \phi$ et $u_{n+2} = 1 + 1/c > \phi$.
Suupposons par exemple que $a = u_0 > \phi$
Alors pour tout $n \in \mathbb{N}$ on a $u_{2n} > \phi$ et $u_{2n+1} < \phi$
On a :
$u_{n+1}-u_n = f(u_n)-f(u_{n-1}) = 1/u_n-1/u_{n-1} = (u_{n-1}-u_n)/(u_{n-1}u_n)$
Comme $(u_{n-1}u_n) > 0$ on en déduit que $u_{n+1} - u_n$ et $u_n - u_{n-1}-$ sont de signe opposé donc que $u_{n+1} - u_n$ et $u_{n-1} - u_{n-2}$ et sont de même signe.
Ainsi $u_{2n} - u_{2n-2}$ a le même signe que $u_2 - u_0$ et
$u_{2n+1} - u_{2n-1}$ a le même signe que $u_3 - u_1$.
Signe de $u_2 - u_0$:
$u_2 - u_0 = (2a + 1)/(a + 1) - a = (-a^2 + a + 1)/(a + 1) < 0$ puisque $a > \phi$ et que ϕ est la plus grande racine de $-x^2 + x + 1$
Signe de $u_3 - u_1$:
$u_3 - u_1 = (2b + 1)/(b + 1) - b = (-b^2 + b + 1)/(b + 1) > 0$ puisque $0 < b < \phi$ et que ϕ est la plus grande racine de $-x^2 + x + 1$ et que 0 se trouve entre les racines.
On a donc montrer que u_{2n} est décroissante et minorée donc est convergente vers la solution positive de $f(f(x)) = x$ qui est ϕ et que u_{2n+1} est croissante et majorée donc est convergente vers la solution positive de $f(f(x)) = x$ qui est ϕ . Donc u_n converge vers ϕ.
Remarque Lorsque $a = 1$, u_n est le quotient de de 2 termes consécutifs de la suite de Fibonacci.

4.9 Deux suites convergentes vers le nombre d'or L

4.9.1 $u_0 = 1$ et $u_{n+1} = \sqrt{1+u_n}$ pour $n \in \mathbb{N}$

Soit la suite u définie par :
$u_0 = 1$ et $u_{n+1} = \sqrt{1+u_n}$ pour $n \in \mathbb{N}$.
Calculer une valeur approché de u_5.
Montrer que u est à terme positif et est croissante (on montrera que $u_{n+1} - u_n > 0$ pour $n \in \mathbb{N}$).
Soit L la racine positive de $x^2 - x - 1 = 0$.
Montrer que $u_n < L$ pour $n \in \mathbb{N}$.
En déduire que u converge vers $L = \dfrac{1+\sqrt{5}}{2}$
On tape :
`sqrt(1+sqrt(1+sqrt(1+sqrt(1+sqrt(2.)))))`
On obtient :
`1.61612120651`
ou bien, on tape :
```
1
sqrt(1.+ans())
```
On valide la dernière ligne 5 fois et on obtient :
`1.61612120651`
ou bien, on utilise le tableur...
On tape :
`mult_conjugate(sqrt(1+u1)-sqrt(1+u0))`
On obtient apres simplification du numérateur :
`(-u0+u1)/(sqrt(1+u1)+sqrt(1+u0))`
ce qui veut dire que $u_2 - u_1$ a la même signe que $u_1 - u_0 = \sqrt{2} - 1 > 0$.
Le même calcul montre que $u_{n+1} - u_n$ a la même signe que $u_n - u_{n-1}$ qui a la même signe que $u_{n-1} - u_{n-2}$ etc..qui a la même signe que $u_1 - u_0$.
Donc $u_{n+1} - u_n > 0$ pour $n \in \mathbb{N}$. La suite u est donc croissante.
On tape :
`solve(x^2-x-1)`
On obtient :
`[1/2*(1-(sqrt(5))),1/2*(1+sqrt(5))]`
Donc $L = 1/2 * (1+\sqrt{5}) \simeq 1.61803398875$ et $L = \sqrt{1+L}$.
On tape :
`mult_conjugate(sqrt(1+un)-sqrt(1+L))`
On obtient :
`(-L+un)/(sqrt(1+un)+sqrt(1+L))`
Donc $u_{n+1} - L$ a le même signe que $u_n - L$ qui a la même signe que $u_{n-1} - L$ etc..qui a la même signe que $u_0 - L = 1/2*(1-sqrt(5)) < 0$. Donc u est majorée par L.
La suite u est croissante et major'e donc u est convergente et sa limite a vérifie $a = \sqrt{1+a}$. Donc $a = L$.
La convergence de u_n n'est pas très rapide puisque u_5 donne la valeur de L avec seulement 2 décimales exactes.

4.9.2 $u_0 = 2$ et $u_{n+1} = \frac{u_n^2+1}{2u_n-1}$ pour $n \in \mathbb{N}$

Soit la suite u définie par :
$u_0 = 2$ et $u_{n+1} = \frac{u_n^2+1}{2u_n-1}$ pour $n \in \mathbb{N}$.
En considérant la fonction g définie par $g(x) = x^2 - x - 1$, montrer que u_n est la suite de la méthode de Newton pour trouver une valeur approchée de `L:=1/2*(1+sqrt(5))` qui est le zéro de g dans $[1, 2]$, c'est à dire que $u_{n+1} = u_n - \frac{g(u_n}{g'(u_n)}$.
Montrer que $u_{n+1} - L = \frac{(u_n-L)^2}{2u_n-1}$.
Montrer par récurrence que $u_n > L$ pour tout $n \in \mathbb{N}$
Montrer que u est décroissante et converge vers L
Calculer une valeur approché de u_5.
On tape :

```
g(x):=x^2-x-1
f(x):=x-g(x)/g'(x)
normal(f(x))
```

On obtient : `(x^2+1)/(2*x-1)`
Donc $u_{n+1} = f(u_n)$.
On tape :

```
L:=1/2*(1+sqrt(5))
normal(f(L))
```

On obtient : `1/2*(1+sqrt(5))`
Donc $f(L) = L$ On tape pour calculer $u_{n+1} - L$:

```
factor(f(un)-f(L))
```

On obtient :

```
(2*(un+(-(sqrt(5))-1)/2)^2)/(4*un-2)
```

On a $u_0 = 2 > L$ et si $u_n > L$ alors $2u_n - 1 > 2L - 1 > 0$ donc puisque $u_{n+1} - L = \frac{(u_n-L)^2}{2u_n-1} > 0$ on en déduit que $u_{n+1} > L$.
On tape pour avoir le signe de $u_{n+1} - u_n$:

```
normal(f(un)-un)
```

On obtient : `(-un^2+un+1)/(2*un-1)`
Comme $u_n > L$ u_n est à l'extérieur des racines de $-x^2+x+1$ donc $-u_n^2+u_n+1 < 0$ et $2u_n - 1 > 0$ donc $u_{n+1} - u_n < 0$ donc u est décroissante et converge vers a la racine positive de $f(x) - x$.
On tape :

```
normal(solve(f(x)-x))
```

On obtient : `[(-(sqrt(5))+1)/2,(sqrt(5)+1)/2]`
Donc $a = L = (\sqrt{5} + 1)/2$
On tape :

```
(f@@5)(2.)
```

On obtient (avec 30 chiffres significatifs) :

```
1.61803398874989484820458683838
```

On tape :

```
(f@@5)(2.)-(sqrt(5)+1)/2.
```

On obtient :

```
0.397270351869301545205346505476 4e-26
```

La convergence de u_n est tres rapide puisque u_5 donne la valeur de L avec 26 décimales exactes.

4.10 Le nombre d'or et $\cos(\frac{\pi}{5})$

4.10.1 Calcul de $\cos(\frac{2\pi}{5})$

Soit $z_1 = \exp(i * \dfrac{2\pi}{5})$.
$z_1 = a + ib = \cos(\dfrac{2\pi}{5}) + i * \sin(\dfrac{2\pi}{5})$ est la racine de $z^5 - 1 = 0$ qui vérifie $a > 0$ et $b > 0$.
Puisque $z^5 - 1 = (z-1)(z^4 + z^3 + z^2 + z + 1)$, $z_1 = a + ib$ est la racine de $z^4 + z^3 + z^2 + z + 1 = (z + \frac{1}{z})^2 + (z + \frac{1}{z}) - 1 = 0$.
On pose $Z = z + \dfrac{1}{z}$ et on tape :
`solve(Z^2+Z-1,Z)`
On obtient : `[1/2*(-1-sqrt(5)),1/2*(-1+sqrt(5))]`
Comme $z_1 = a + ib$ est de module 1, on a $\frac{1}{z1} = a - ib$ et donc $z1 + \frac{1}{z1} = 2a = 2\cos(\frac{2\pi}{5})$.
On a donc :

$$\cos(\frac{2\pi}{5}) = \frac{-1+\sqrt{5}}{4}$$

On tape :
`normal(expand((1/4*(-1+sqrt(5)))^2))`
On obtient : `(-sqrt(5)+3)/8`
Donc :

$$\cos(\frac{2\pi}{5})^2 = \frac{3-\sqrt{5}}{8}$$

$$\sin(\frac{2\pi}{5})^2 = \frac{5\sqrt{5}}{8}$$

4.10.2 Calcul de $2\cos(\frac{\pi}{5})$

On a : $\cos(\dfrac{2\pi}{5}) = 2\cos(\dfrac{\pi}{5})^2 - 1$
Donc :
$2\cos(\frac{\pi}{5})^2 = \cos(\frac{2\pi}{5}) + 1) = \frac{-1+\sqrt{5}}{4} + 1$
On tape : `normal(1/4*(-1+sqrt(5))+1)`
On obtient : `(sqrt(5)+3)/4`
Donc :

$$\cos(\frac{\pi}{5})^2 = \frac{3-\sqrt{5}}{8}$$

On tape : `normal(expand((1/4*(1+sqrt(5)))^2))`
On obtient : `(sqrt(5)+3)/8`
Donc :

$$\cos(\frac{\pi}{5}) = \frac{1+\sqrt{5}}{4}$$

Le nombre d'or est :

$$\phi = \frac{1+\sqrt{5}}{2} \simeq 1.61803398875$$

Donc

$$2\cos(\frac{\pi}{5}) = \phi$$

4.10.3 Les décimales de $\frac{10}{89}$ et la suite de Fibonacci

Le constat est le suivant :
On tape le programme donnant les n premiers termes de la suite de Fibonacci :

```
Fibonasuite(n):={
local a,b,c,L,k;
L:=NULL;
si n<=0 alors retourne L;fsi;
L:=L,1;
si n==1 alors retourne L;fsi;
L:=L,1;
si n==2 alors retourne L; fsi;
a:=1;
b:=1;
pour k de 3 jusque n faire
    c:=a+b;
    L:=L,c;
    a:=b;
    b:=c;
fpour;
retourne L;
}:;
```

On tape :
`Fibonasuite(11)`
On obtient la suite des 11 premiers termes de la suite de Fibonnacci :
`1,1,2,3,5,8,13,21,34,55,89`
On tape :
`evalf(10/89,50)`
On obtient :
`0.11235955056179775280898876404494382022471910112360`
On tape :
`L:=Fibonasuite(50):;`
On obtient :
`"Done"`
`evalf(sum(L[k]/10^(k+1),k=0..49),50)`
On obtient :
`0.11235955056179775280898876404494382022469480490115`
On a légalité des 39 premières décimales Mais avec : `L:=Fibonasuite(64):;`
et
`evalf(sum(L[k]/10^(k+1),k=0..63),50)`
On obtient légalité des 49 premières décimales :
`0.11235955056179775280898876404494382022471910112360`
Question
Si u_n est la suite de Fibonacci, a-t-on :
$\sum n = 0^{+\infty} \frac{u_n}{10^{n+1}}$ vaut $\frac{10}{89}$?
Qu'a donc 89 de si particulier ? ? ? ? ? La particularité de 89 est que :
$89 = 100 - 10 - 1 = 10^2 - 10^1 - 10^0$ On va montrer que si u_n est la suite de

Fibonacci, on a :
$S = \sum k = 0^{+\infty} \frac{89u_k}{10^k}$ vaut 100.
Pour cela en écrivant $89 = 10^2 - 10^1 - 10^0$, on a :
$S = \sum k = 0^{+\infty} \frac{u_k}{10^{k-2}} - \sum k = 0^{+\infty} \frac{u_k}{10^{k-1}} - \sum k = 0^{+\infty} \frac{u_k}{10^k}$
On a en posant $j = k$:
$S1 = \sum k = 0^{+\infty} \frac{u_k}{10^{k-2}} = 100u_0 + 10u_1 + \sum j = 2^{+\infty} \frac{u_j}{10^{j-2}}$
En posant $j = k + 1$ on a :
$S_2 = \sum k = 0^{+\infty} \frac{u_k}{10^{k-1}} = \sum j = 1^{+\infty} \frac{u_{j-1}}{10^{j-2}} = 10u_0 + \sum j = 2^{+\infty} \frac{u_{j-1}}{10^{j-2}}$
et en posant $j = k + 2$ on a :
$S_3 = \sum k = 0^{+\infty} \frac{u_k}{10^k} = \sum j = 2^{+\infty} \frac{u_{j-2}}{10^{j-2}}$
Donc :
$S = 100u_0 + 10u_1 - 10u_0 + \sum j = 2^{+\infty} \frac{u_j - u_{j-1} - u_{j-2}}{10^{j-2}}$ on sait que :
$u_0 = u_1 = 1$ et $u_j = u_{j-1} + u_{j-2}$ pour $j \geq 2$ donc :
$S = \sum k = 0^{+\infty} \frac{89u_k}{10^k} = 100$ Ce qui implique que, si u_n est la suite de Fibonacci, on a :

$$\frac{10}{89} = \sum k = 0^{+\infty} \frac{u_k}{10^{k+1}}$$

Variante Soit la suite v_n définie par $v_0 = 1, v_1 = 1$ et par $v_{n+2} = 2v_{n+1} + v_n$ (≥ 0).
On a :
$\sum j = 2^{+\infty} \frac{v_j - 2v_{j-1} - v_{j-2}}{10^{j-2}} = 0$ $79 = 100 - 20 - 1$ Soit :
$S = \sum k = 0^{+\infty} \frac{79u_k}{10^k} = \sum k = 0^{+\infty} \frac{(100-20-1)u_k}{10^k}$ $S = 100v_0 + 10v_1 + \sum k = 2^{+\infty} \frac{v_k}{10^{k-2}} - 20v_0 - 2\sum k = 2^{+\infty} \frac{v_{k-1}}{10^{k-2}} - \sum k = 2^{+\infty} \frac{v_{k-2}}{10^{k-2}}$
Donc $S = 100v_0 + 10v_1 - 20v_0 = 90$ On a donc :

$$\frac{9}{79} = \sum k = 0^{+\infty} \frac{v_k}{10^{k+1}}$$

Soit la suite w_n définie par $w_0 = 1, w_1 = 2$ et par $w_{n+2} = 2w_{n+1} + w_n$ (≥ 0).
On a alors :
$S = 100w_0 + 10w_1 - 20w_0 = 100$ On a donc :

$$\frac{10}{79} = \sum k = 0^{+\infty} \frac{w_k}{10^{k+1}}$$

On tape :

```
Suiterec(n,p):={
local a,b,c,L,k;
L:=NULL;
si n<=0 alors retourne L;fsi;
L:=L,1;
si n==1 alors retourne L;fsi;
L:=L,p;
si n==2 alors retourne L; fsi;
a:=1;
b:=p;
pour k de 3 jusque n faire
    c:=a+p*b;
    L:=L,c;
```

```
    a:=b;
    b:=c;
fpour;
retourne L;
}:;
```

Alors si :
`L:=Suiterec(100,p)` on a avec 48 décimales légalité :

$$\frac{10}{69} \simeq \sum k = 0^{99} \frac{L[k]}{10^{k+1}}$$

4.11 Un jeu avec une variante de la suite de Fibonnacci

Ce jeu est décrit par R.V.Heath.
Vous faites des additions plus vite qu'une calculatrice : la preuve, vous demandez à votre interlocuteur de choisir , à votre insu, 2 nombres a et b, puis d'écrire dans une colonne les 10 premiers termes de la suite V définie par :
$V_1 = a$, $V_2 = b$, $V_{n+1} = V_n + V_{n-1}$ pour $n \geq 2$.
Par exemple :
pour $a = 8$ et $b = 7$, on obtient :
8
7
12
22
37
59
96
155
251
406
Quand il a fini, il vous montre les 10 premiers termes de la suite V et vous êtes capable instantanément de donner la somme de ses 10 nombres.
La clé : il vous suffit de multiplier V_7 (le 4ième nombre en partant de la fin) par 11 ce qui donne ici 96*11=1056.
On vérifie avec `Xcas`, on tape :

```
V(a,b,n):={
si n==1 alors retourne a fsi;
si n==2 alors retourne b fsi;
retourne V(b,a+b,n-1);
}:;
```

`L:=V(8,7,k)$(k=1..10)`
On obtient :
`8,7,15,22,37,59,96,155,251,406`
On tape :
`sum(L),V(8,7,7)*11`
On obtient :
`1056,1056`

Pourquoi cela ?
On va montrer que $\sum_{k=1}^{10} V_k = 11V_7$.
Soit F_n la suite de Fibonnacci définie pour $n \geq 0$ par :
$F_0 = 0$, $F_1 = 1$, $F_2 = 1$, $F_{n+1} = F_n + F_{n-1}$ pour $n \geq 2$.

Calcul de $S_n = \sum_{k=0}^{n} F_k$
Les premiers termes de S_n pour $n \geq 0$ sont :
$0, 1, 2, 4, 7, 12, 20, 33, 54, 88, ...$
et les premiers termes de F_n pour $n \geq 0$ sont :
$0, 1, 1, 2, 3, \ 5, \ 8, 13, 21, 34, 55, 89...$
On suppose que :
$S_n = F_{n+2} - 1$
On montre par récurrence que l'on a $S_n = F_{n+2} - 1$:
si pour tout $0 \leq k \leq n$ on a $S_k = F_{k+2} - 1$ alors pour $0 \leq k = n + 1$ on a $S_{n+1} = S_n + F_{n+1} = F_{n+2} - 1 + F_{n+1} = F_{n+3} - 1$ donc
$S_k = F_{k+2} - 1$ pour $k \geq 0$

Calcul de V_k en fonction de F_k pour $k \geq 1$
On a :
$V_1 = a$,
$V_2 = b$,
$V_3 = V_1 + V_2 = a + b$
$V_4 = V_2 + V_3 = a + 2b$
$V_5 = V_3 + V_4 = 2a + 3b$
$V_6 = V_4 + V_5 = 3a + 5b$
$V_7 = V_5 + V_6 = 5a + 8b$
On voit que les coefficients α_n de a sont :
$1, 0, 1, 1, 2, 3...$ i.e $\alpha_n = F_{n-2}$ pour $n \geq 2$ et
les coefficients β_n de b sont :
$0, 1, 1, 2, 3, 5...$ i.e $\beta_n = F_{n-1}$ pour $n \geq 1$ et
On montre par récurrence que l'on a :
$V_k = aF_{k-2} + bF_{k-1}$ pour $k \geq 2$
si pour tout $2 \leq k \leq n$ on a $V_k = aF_{k-2} + bF_{k-1}$ alors pour $2 \leq k \leq n$ on a $V_{n+1} = aF_{n-2} + bF_{n-1} + aF_{n-3} + bF_{n-2}$ donc
$V_{n+1} = a(F_{n-2} + F_{n-3}) + b(F_{n-1} + F_{n-2}) = aF_{n-1} + bF_n$. c.q.f.d.

Calcul de $\sum_{k=1}^{10} V_k$
$\sum_{k=1}^{n} V_k = a + b + a\sum_{k=3}^{n} F_{k-2} + b\sum_{k=3}^{n} F_{k-1}$
$\sum_{k=1}^{n} V_k = a + b + a\sum_{k=1}^{n-2} F_k + b\sum_{k=2}^{n-1} F_k$
$\sum_{k=1}^{n} V_k = a + b + aS_{n-2} + bS_{n-1} - b = aF_n + bF_{n+1} - b$.

Que vaut $\sum_{k=1}^{10} V_k$
$\sum_{k=1}^{10} V_k = aF_{10} + bF_{11} - b = 55a + 88b$
Que vaut V_7
$V_7 = 5a + 8b$
et on bien !
$\sum_{k=1}^{10} V_k = 55a + 88b = 11(5a + 8b) = 11V_7$

Chapitre 5

Les fractions continues et π

5.1 Développement en fractions continues de π

Avec `Xcas` si on tape :
`evalf(pi,21)`
On obtient :
`3.141592653589793238462`
Une valeur approchée par excès de π par un rationnel est :

$$\frac{22}{7}$$

On a $\frac{22}{7} \simeq 3.14285714286$ ce qui fait 2 décimales exactes et
`evalf(22/7-pi)` renvoie `0.00126448926721`
Donc $0 < \frac{22}{7} - 0.0013 < \pi < \frac{22}{7}$ et $0 < \frac{22}{7} - \pi < 1.3 10^{-3}$

Une autre valeur approchée par excès de π par un rationnel est :

$$\frac{355}{113}$$

On a $\frac{355}{113} \simeq 3.14159292035$ ce qui fait 6 décimales exactes et
`evalf(355/113-pi)` renvoie `2.66764118351e-07`
Donc $0 < \frac{355}{113} - 2.67e - 07 < \pi < \frac{355}{113}$
Une valeur approchée par défaut de π par un rationnel est :

$$\frac{333}{106}$$

On a $\frac{333}{106} \simeq 3.14150943396$ ce qui fait 4 décimales exactes et
`evalf(pi-333/106)` renvoie `8.32196276406e-05`
Donc

$$\frac{333}{106} < \pi < \frac{333}{106} + 8.33e - 05$$

Ces approximations proviennent du développement en fractions continues de π.
On tape :
`dfc(pi)`
On obtient :
`[3,7,15,1,292,1,1,1,2]`

On tape :
```
dfc2f([3,7])
```
On obtient $3+1/7$:
```
22/7
```
On tape :
```
dfc2f([3,7,15])
```
On obtient $3+1/(7+1/15)$:
```
333/106
```
On tape :
```
dfc2f([3,7,15,1])
```
On obtient $3+1/(7+1/(15+1))$:
```
355/113
```
On tape :
```
dfc2f([3,7,15,1,292])
```
On obtient $3+1/(7+1/(15+1/(1+1/292)))$:
```
103993/33102
```
On tape :
```
dfc2f([3,7,15,1,292,1])
```
On obtient $3+1/(7+1/(15+1/(1+1/(292+1))))$:
```
104348/33215
```
On tape :
```
dfc2f([3,7,15,1,292,1,1])
```
On obtient $3+1/(7+1/(15+1/(1+1/(292+1/(1+1))))))$:
```
208341/66317
```
On tape :
```
dfc2f([3,7,15,1,292,1,1,1])
```
On obtient $3+1/(7+1/(15+1/(1+1/(292+1/(1+1/(1+1))))))$:
```
312689/99532
```
On tape :
```
sort([22/7,355/113,103993/33102,104348/33215,208341/66317,312689/99532
```
On obtient :
```
[103993/33102,208341/66317,pi,312689/99532,104348/33215,355/113,22/7]
```

5.2 Exercice

Montrer les formules :

$$\int_0^1 \frac{x^4(1-x)^4}{1+x^2}dx = \frac{22}{7} - \pi$$

$$\int_0^1 \frac{x^8(1-x)^8(25+816x^2)}{3164(1+x^2)}dx = \frac{355}{113} - \pi$$

5.3 construction d'un carré ayant pour aire $\frac{22}{7}$

Ce carré a pour coté : $\frac{\sqrt{22}}{\sqrt{7}}$.
On a :
$22 = 5^2 - 3$ et $7 = 2^2 + 3$

Donc :
$\sqrt{22}$ est le 2ième côté de l'angle droit d'un triangle rectangle dont l'hypoténuse a pour longueur 5 et un côté de l'angle droit a pour longueur $\sqrt{3}$.
$\sqrt{7}$ est la longueur de l'hypoténuse d'un triangle rectangle dont les côtés de l'angle droit ont pour longueur 2 et $\sqrt{3}$.
On tape :

```
triangle_equilateral(1,3,C):;
C:=C;
A:=point(-1);B:=point(1);
l:=normal(longueur(A,C));
D:=point(4);
segment(A,D);
c:=cercle(A,D);
F:=inter(c,cercle(A,l/2),C);
normal(longueur2(F,D));
normal(longueur2(D,C));
G:=D+(C-D)/longueur(C,D);
normal(longueur2(D,G));
H:=normal(inter_unique(segment(D,F),parallele(G,droite(F,C))));
normal(longueur(H,D));
segment(F,D),segment(D,C);
carre(D,H,K,L,affichage=1);
```

On obtient l'aire du carré rouge approche par excèsà $1.3 10^{-3}$ l'aire du cercle rouge de diamètre AB :



On a :
`normal(longueur(A,C));` renvoie `2*sqrt(3)`

`normal(longueur2(F,D)` renvoie `22`
`normal(longueur2(D,C));` renvoie `7`
`normal(longueur2(D,G))` renvoie `1`
`normal(longueur(H,D));` renvoie `(sqrt(154))/7`
`normal(longueur2(H,D));` renvoie `22/7`
`aire(carre(D,H))` renvoie `22/7`

5.4 Construction par Ramanujan d'un carré d'aire $\frac{355}{113}$

En 1913 Ramanujan proposa la construction d'un carré ayant pour aire celle d'un cercle de rayon 1 avec une erreur de $1.47 * 10^{-5}$.
Ce carré a pour coté : $\frac{\sqrt{355}}{\sqrt{113}}$.
On a :
$355 = 18^2 + 31$ et $113 = 12^2 - 31$
$\frac{355}{9^2} = 4 + \frac{31}{9^2}$ et $\frac{113}{6^2} = 4 - \frac{31}{6^2}$
Il reste a construire $\sqrt{31}$.
On a :
$31 = 6^2 - 5$
Voici la construction de ce carré faite par Ramanujan.
Soit le cercle c de centre O et de diamètre $PR = 2$.
Avec `Xcas`, on prend :
`O:=point(0);P:=point(-1);R:=point(1);`
`H:=point(-1/2);T:=point(2/3);`
Le cercle c passe par le point `Q:=point(2/3+i*sqrt(5)/3);`
On a `RS=QT=sqrt(5)/3`
On définit les projections M et N respectives de O et T sur PS.
On définit :
`L:=point(-1-i*longueur(M,N));`
`K` sur le cercle c tel que $PK = PM$
`C` sur le segment RK tel que $RC = RH = 3/2$ `D` sur le segment RL tel que $CD//KL$
Le carré de côté RD a alors pour côté :

$$\sqrt{\frac{355}{113}}$$

Avec `Xcas`, ontape :

```
O:=point(0);
P:=point(-1);
R:=point(1);
H:=point(-1/2);
T:=point(2/3);
c:=cercle(O,1,affichage=4+epaisseur_ligne_2);
Q:=point(2/3+i*sqrt(5)/3);
S:=inter_unique(c,cercle(R,sqrt(5)/3),Q);
s:=segment(P,S);
M:=projection(s,O);
N:=projection(s,T);
```

```
L:=point(-1-i*longueur(M,N));
K:=inter_unique(c,cercle(P,longueur(P,M)),L);
C:=inter_unique(droite(R,K),cercle(R,3/2),K);
D:=inter_unique(droite(L,R),parallele(C,droite(L,K)));
carre(D,R,affichage=1+epaisseur_ligne_2);
segment(P,L);
segment(R,S);
segment(R,L);
segment(R,K);
segment(L,K);
segment(D,C);
segment(O,M);
segment(T,N);
cercle(P,longueur(M,P),affichage=ligne_tiret_point);
cercle(R,3/2,affichage=ligne_tiret_point);
```

On tape :
`evalf(aire(carre(D,R)))`
On obtient :
`3.14157798199`
On tape :
`evalf(pi-aire(carre(D,R)))`
On obtient :
`1.46716034237e-05`

On obtient la figure :



Calcul de RD^2
On a :
$RS^2 = 5/9$
$PS^2 = 4 - 5/9 = 31/9$
$PK^2 = PM^2 = PS^2/4 = 31/36$
$PL^2 = MN^2 = PS^2/9 = 31/81$
$RK^2 = 4 - PK^2 = 113/36$
$RL^2 = 4 + PL^2 = 355/81$
$RC^2 = RH^2 = 9/4$
$\frac{RD^2}{RL^2} = \frac{R^2C}{RK^2} = \frac{81}{113}$
Donc $RD^2 = \frac{355}{113}$

5.5 Une valeur approchée de π par $(9^2 + 19^2/22)^(1/4)$

On tape :
```
A:=(9^2+19^2/22)^(1/4)
evalf(A)
```
On obtient :
```
3.14159265258
```
On tape : `evalf(pi)`
On obtient : `3.14159265359`
On tape :
```
evalf(pi-A)
```
On obtient :
```
1.00715169538e-09
```

5.6 Un nombre quasi-entier : $(\frac{\ln(5280^3*(236674+30303*\sqrt{61})^3+744)}{\pi})^2$

Avec `Xcas`, on tape :
```
A:=ln(5280^3*((236674+30303*sqrt(61))^(3))+744)
evalf((A/pi)^2)
```
On obtient :
```
427.0
```
On tape :
```
evalf((A/pi)^2,52)
```
On obtient :
```
0.4270000000000000000000000000000000000000000000000000e3
```
On tape :
```
evalf((A/pi)^2,55)
```
On obtient :
```
0.4270000000000000000000000000000000000000000000000000106e3
```

Chapitre 6

Pour trouver les premières décimales de π

6.1 La formule de Gregory et le développement en série de $\arctan(x)$

La série de Gregory est le développement en série entière de $\arctan(x)$
On a :

$$\arctan(x) = x - \frac{x^3}{3} + \frac{x^5}{5} + ... + (-1)^k \frac{x^{2k+1}}{2k+1} + ...$$

Le reste de cette série alternée est du signe du premier terme négligé et est majoré en valeur absolue par la valeur absolue du premier terme négligé.

6.2 La formule de Machin

De la formule d'addition des tangentes à savoir :

$$\tan(a+b) = \frac{\tan(a) + \tan(b)}{1 - \tan(a) * \tan(b)}$$

on en déduit la formule pour $ab < 1$:

$$\arctan(a) + \arctan(b) = \arctan(\frac{a+b}{1-ab})$$

Exercices :
1/ Pour $a = 1/5$ on cherche b pour que :

$$\arctan(a) + \arctan(b) = \arctan(1) = \frac{\pi}{4}$$

On doit donc résoudre :

$$\frac{a+b}{1-ab} = 1$$

On trouve :

$$b = \frac{1-a}{1+a} = \frac{2}{3}$$

Donc :

$$\arctan(\frac{1}{5}) + \arctan(\frac{2}{3}) = \frac{\pi}{4}$$

2/ Pour $a = 1/5$ on cherche b pour que :

$$\arctan(a) + \arctan(b) = \arctan(\frac{2}{3})$$

On doit donc résoudre :

$$\frac{a+b}{1-ab} = c = \frac{2}{3}$$

On trouve :

$$b = \frac{c-a}{1+ac} = \frac{\frac{2}{3}-\frac{1}{5}}{1\frac{2}{15}} = \frac{7}{17}$$

Donc :

$$2\arctan(\frac{1}{5}) + \arctan(\frac{7}{17}) = \frac{\pi}{4}$$

3/ En faisant encore deux fois le même genre de substitution, montrer la formule de Machin :

$$4\arctan(\frac{1}{5}) - \arctan(\frac{1}{239}) = \frac{\pi}{4}$$

Pour cela on prend successivement :
$c = \frac{7}{17}$ et on trouve $b = \frac{c-a}{1+ac} = \frac{9}{46}$ et
$c = \frac{9}{46}$ et on trouve $b = \frac{c-a}{1+ac} = \frac{-1}{239}$
4/ Écrire un programme qui prend en entrée a et n et qui renvoie la liste L des valeurs de b_k vérifiant :
$k\arctan(a) + \arctan(b_k) = \frac{\pi}{4}$ pour $k = 1..n$ ($L[k-1] = b_k$)
On tape le programme :

```
machin1(a,n):={
local k,c,L;
c:=1;
L:=[0];
for (k:=1;k<n+1;k:=k+1) {
c:=(c-a)/(1+a*c);
L:=append(L,c);
}
return(L);
};
```

On tape :
`machin1(1/5,5)`
On obtient :
`[0,2/3,7/17,9/46,1/-239,-122/597]`
Ainsi, pour $a = 1/5$ et $k = 3$, $b_k = 9/46$ donc :
$3\arctan(\frac{1}{5}) + \arctan(\frac{9}{46}) = \frac{\pi}{4}$
On tape :
`machin1(1/3,5)`
On obtient :
`[0,1/2,1/7,-2/11,-17/31,-41/38]`

Ainsi, pour $a = 1/3$ et $k = 2$, $b_k = 1/7$ donc :
$2\arctan(\frac{1}{3}) + \arctan(\frac{1}{7}) = \frac{\pi}{4}$

6.3 Les décimales de π avec les formules précédentes

6.3.1 Une remarque

Si on utilise pour calculer π la formule :

$$\arctan(x) = x - \frac{x^3}{3} + \frac{x^5}{5} + ... + (-1)^k \frac{x^{2k+1}}{2k+1} + ...$$

Si on prend $x = 1$, on a $\arctan(1) = \pi/4 = 1 - \frac{1}{3} + \frac{1}{5} + ... + (-1)^k \frac{1}{2k+1} + ...$
mais la convergence est lente.
On remarque que la convergence est beaucoup plus rapide pour $x = 1/5$ et encore plus rapide pour $x = 1/239$ d'où l'utilisation de la formule :

$$\frac{\pi}{4} = 4\arctan(\frac{1}{5}) - \arctan(\frac{1}{239})$$

6.3.2 Le programme avec `Xcas`

Pour calculer la somme de n termes de la série on va utiliser la méthode de Hörner pour faire le moins possibles de multiplications, on a :
$\arctan(x) = x(1 - x^2(\frac{1}{3} - x^2(\frac{1}{5} - x^2(.... - x^2(\frac{1}{2n-1})))))$
L'utilisateur doit rentrer la valeur de x (a) et le nombre n de termes de la série.

```
//approx de arctan(a) par sa serie :
// a-a^3/3+..+(-1)^n*a^(2n+1)/(2n+1)
//cette valeur est calculee par la methode de Horner
gregory(a,n):={
local t,k;
t:=1/(2*n-1);
for (k:=2*n-3;k>0;k:=k-2) {
t:=1/k-a^2*t;
}
return (a*t);
};
```

On tape :
`16*gregory(1/5,18)-4*gregory(1/239,6)`
On obtient :
`3.14159265359` On tape :
`evalf(16*gregory(1/5,42)-4*gregory(1/239,20))`
On obtient, si on a choisit 60 `Chiffres` dans la configuration du CAS (menu `Cfg->Configuration du CAS`):
`3.14159265358979323846264338327950288419716939937510582097494`

6.3.3 Combien faut-il calculer de termes ?

Soit $R_n(x)$ le reste de la série $\sum_{k=1}^{\infty}(-1)^{k+1}\frac{x^{2k-1}}{2k-1}$: $R_n(x)=\sum_{k=n+1}^{\infty}(-1)^{k+1}\frac{x^{2k-1}}{2k-1}$.
On sait que $|R_n(x)|<\frac{|x|^{2n+1}}{2n+1}$ Pour avoir $|R_n(1/5)|=|R_n(0.2)|<10^{-61}$ il faut que :
$2^{2n+1}<(2n+1)10^{2n-60}$ et comme $2^10\simeq 10^3$ cela donne si on suppose $2n+1>10$:
$10^{(6n+3)/10}<10^{2n-59}$ soit $593<14n$ soit $n\simeq 42$ On vérifie pour $n=42$ on a $\frac{(1/5)^{85}}{85}<2.56e-62$
On peut aussi écrire si on suppose que $2n+1>10$:
$\frac{|x|^{2n+1}}{2n+1}<x^{2n+1}/10<10^{-61}$ donc on va choisir $x^{2n+1}<10^{-60}$ soit $(2n+1)log10(x)<-60$ ou encore $n>((-60)/log10(x)-1)/2$ Pour $x=1/5$ on a $n>42.4202967422$ et comme $2n+1>40$ on peut améliorer la majoration $\frac{|x|^{2n+1}}{2n+1}<x^{2n+1}/40<10^{-61}$
ce qui donne $n>((-60+log10(4))/log10(1/5)-1)/2=41.9896201841$ donc on prend $n=42$
pour $x=1/293$ on a $n>11.66$ donc on prend $n=12$ et on vérifie : $\frac{(1/239)^{25}}{25}=1.38711499837e-61$.
On choisit 62 `Chiffres` dans la configuration du CAS (menu `Cfg->Configuration du CAS`) et on tape :
`evalf(16*gregory(1/5,42)-4*gregory(1/239,12))`
On obtient :
`3.14159265358979323846264338327950288419716939937510582097494446`
On tape :
`evalf(pi)`
On obtient :
`3.14159265358979323846264338327950288419716939937510582097494446`
Remarque
Avec cette méthode John Machin calcula 100 décimales de π en 1706.

6.3.4 Les formules de même type que celles de Machin

En 1973, Jean Guilloud a mis une journée pour calculer 10^6 décimales de π en utilisant une formule de même type à savoir :

$$6\arctan(\frac{1}{8})+2\arctan(\frac{1}{57})+\arctan(\frac{1}{239})=\frac{\pi}{4}$$

en vérifiant ses calculs avec la formule analogue :

$$12\arctan(\frac{1}{18})+8\arctan(\frac{1}{57})-5\arctan(\frac{1}{239})=\frac{\pi}{4}$$

En 1999, Yasumata Kanadaa a atteint le record en calculant $12411*10^8$ décimales de π en utilisant une formule de même type à savoir :

$$24\arctan(\frac{1}{12943})-12\arctan(\frac{1}{682})+44\arctan(\frac{1}{57})+7\arctan(\frac{1}{239})=\frac{\pi}{4}$$

et la formule analogue :

$$12\arctan(\frac{1}{49})+32\arctan(\frac{1}{57})-5\arctan(\frac{1}{239})+12\arctan(\frac{1}{110443})=\frac{\pi}{4}$$

On peut vérifier ces formules avec `Xcas`, on tape par exemple :
`tsimplify(12*atan(1/49)+32*atan(1/57)-5*atan(1/239)+12*atan(1/110443))`
On obtient :
$\frac{\pi}{4}$
Comment trouver des formules de type Machin ?
voir aussi 7.5
Montrons pour cela que si $a \in \mathbb{N}^*$ et si $a^2+1 = a_1 * a_2$ avec $(a_1, a_2) \in \mathbb{Z}^2$ et $(a+a_1)(a+a-2)\neg 0$ alors on a :
$\arctan(1/a) = \arctan(1/(a+a_1)) + \arctan(1/(a+a_2))$ (si $a \geq 2$ alors a ne divise pas a^2+1 donc $a+a-1$ et $a+a_2$ sont non nuls et différents).
On a si $xy < 1$, $\arctan(x) + \arctan(y) = \arctan((x+y)/(1-xy))$ donc $\arctan(1/(a+a_1)) + \arctan(1/(a+a_2)) = \arctan((2a+a_1+a_2)/((a+a_1)(a+a_2)-1)) = \arctan(1/a)$ puisque

— si $a+a_1 > 1$ et $a+a_2 > 1$ alors $1/(a+a_1) * 1/(a+a_2) < 1$
— si $a+a_1 > 1$ et $a+a_2 < -1$ alors $1/(a+a_1) * 1/(a+a_2) < 0 < 1$
— si $a+a_1 \leq -1$ et $a+a_2 < -1$ alors $1/(a+a_1) * 1/(a+a_2) = 1/(-a-a_1) * 1/(-a-a_2) < 1$

Donc :
si $a^2+1 = a_1 * a_2$ alors $\arctan(1/(a+a_1)) + \arctan(1/(a+a_2)) = \arctan(1/a)$
Exemples
si $a = 1$ on a $\arctan(1) = \arctan(1/2) + \arctan(1/3)$.
si $a = 2$ on a $\arctan(1/2) = \arctan(1/3) + \arctan(1/7)$ et
$\arctan(1/2) = \arctan(1/(2-1)) + \arctan(1/(2-5)) = \arctan(1) - \arctan(1/3)$
On a donc :
$a = 1$ $a^2+1 = 2 = 1*2$ donc $\pi/4 = \arctan(1) = \arctan(1/2) + \arctan(1/3)$
$a = 2$ $a^2+1 = 5 = 1*5$ donc $\arctan(1/2) = \arctan(1/3) + \arctan(1/7)$ et
$\arctan(1/2) = \arctan(1/(2-1)) + \arctan(1/(5-2)) = \arctan(1) - \arctan(1/3)$
$a = 3$ $a^2+1 = 10 = 1*10 = 2*5$ donc $\arctan(1/3) = \arctan(1/4) + \arctan(1/13) = \arctan(1/5) + \arctan(1/8)$ et $\arctan(1/3) = \arctan(1/(3-2)) + \arctan(1/(3-5)) = \arctan(1) - \arctan(1/2)$
$a = 5$ $a^2+1 = 26 = 1*26 = 2*13$ donc $\arctan(1/5) = \arctan(1/7) + \arctan(1/18) = \arctan(1/6) + \arctan(1/31)$ et
$\arctan(1/5) = \arctan(1/(5-2)) + \arctan(1/(5-13)) = \arctan(1/3) - \arctan(1/8)$ $a = 7$ $a^2+1 = 50 = 1*50 = 2*25$ donc $\arctan(1/7) = \arctan(1/8) + \arctan(1/57) = \arctan(1/9) + \arctan(1/32)$ et
$\arctan(1/7) = \arctan(1/(7-2)) + \arctan(1/(7-25)) = \arctan(1/5) - \arctan(1/18)$

On en déduit donc que :

$$\pi/4 = 2\arctan(1/3) + \arctan(1/7)$$

$$\pi/4 = 2\arctan(1/3) + \arctan(1/5) - \arctan(1/18)$$

$$\pi/4 = 2\arctan(1/3) + \arctan(1/5) - \arctan(1/18)$$

$$\pi/4 = 2\arctan(1/3) + \arctan(1/8) + \arctan(1/57)$$

$$\pi/4 = 3\arctan(1/3) - \arctan(1/5) + \arctan(1/57)$$

et en utilisant $\pi/4 = 4\arctan(1/5) - \arctan(1/239)$ on retrouve facilement les 2 formules utilisées par Jean Guilloud :
$6\arctan(\frac{1}{8}) + (6-4)\arctan(\frac{1}{57}) + \arctan(\frac{1}{239}) =$
$6(\pi/4 - 2\arctan(1/3)) - 4(\pi/4 - 3\arctan(1/3) + \arctan(1/5)) - \pi/4 + 4\arctan(1/5) = \pi/4$
et
$12\arctan(\frac{1}{18}) + 8\arctan(\frac{1}{57}) - 5\arctan(\frac{1}{239}) =$
$12(-\pi/4 + 2\arctan(1/3) + \arctan(1/5)) + 8(\pi/4 - 3\arctan(1/3) + \arctan(1/5)) + 5(\pi/4 - 4\arctan(1/5)) = \pi/4$

6.4 Pour bien afficher les décimales

```
//approx de arctan(a) par a-a^3/3+..+(-1)^n*a^(2n+1)/(2n+1)
//valeur de la serie est calculee par la methode de Horner
gregory(a,n):={
local t,k;
t:=1/(2*n-1);
for (k:=2*n-3;k>0;k:=k-2) {
t:=1/k-a^2*t;
}
return (a*t);
}
:;
//pour afficher selon un tableau
affiche1(fl,n):={
local s,p,q,j,k,L,LT;
L:="";
s:=string(fl);
p:=size(s);
q:=iquo(p,11*n)
si irem(p,11*n)==0 alors
LT:=[0$q];sinon
LT:=[0$(q+1)];
fsi;
pour j de 0 jusque q-1 faire
pour k de 1 jusque 11 faire
L:=L+mid(s,0,n)+" ";
s:=mid(s,n);
fpour;
LT[j]=<[L];
L:="";
fpour;
si r!=0 alors
pour k de 1 jusque size(s) faire
L:=L+mid(s,0,n)+" ";
s:=mid(s,n);
fpour;
```

```
LT[q]=<[L];
fsi;
return LT;
}:;
//pour afficher selon une liste de chaines
affiche2(fl,n):={
local s,p,q,j,k,L,LT;
L:="";
s:=string(fl);
p:=size(s);
q:=iquo(p,11*n);
r:=irem(p,11*n);
si r==0 alors
LT:=[0$q];sinon
LT:=[0$q+1];
fsi;
pour j de 0 jusque q-1 faire
pour k de 1 jusque 11 faire
L:=L+mid(s,0,n)+" ";
s:=mid(s,n);
fpour;
LT[j]=<L;
L:="";
fpour;
si r!=0 alors
pour k de 1 jusque size(s) faire
L:=L+mid(s,0,n)+" ";
s:=mid(s,n);
fpour;
LT[q]=<L;
fsi;
return LT;
}:;
```

On vérifie pour $n = 2900$ on a $\frac{(1/5)^{85}}{5801} < 0.33e - 4058$
et on vérifie pour $n = 860$ on a $\frac{(1/239)^{11721}}{1721} = 0.35e - 4096$.
On tape :
`(1/5.)^5801/5801,(1/239.)^1721/1721`
On obtient :
`0.3247147235398264366269e-4058,0.347881711410409416161e-4096`
`Digits:=4000`
On tape :
`LR1:=affiche1(evalf(16*gregory(1/5,2900)-4*gregory(1/239,860)),5)`
On obtient comme réponse une matrice.
Ou on tape :
`LR2:=affiche2(evalf(16*gregory(1/5,2900)-4*gregory(1/239,860)),5)`
`pour j de 0 jusque size(LR2)-1 faire print(LR2[j]) fpour`
On obtient après 0.88s 3993 décimales :

3.141 59265 35897 93238 46264 33832 79502 88419 71693 99375 10582
09749 44592 30781 64062 86208 99862 80348 25342 11706 79821 48086
51328 23066 47093 84460 95505 82231 72535 94081 28481 11745 02841
02701 93852 11055 59644 62294 89549 30381 96442 88109 75665 93344
61284 75648 23378 67831 65271 20190 91456 48566 92346 03486 10454
32664 82133 93607 26024 91412 73724 58700 66063 15588 17488 15209
20962 82925 40917 15364 36789 25903 60011 33053 05488 20466 52138
41469 51941 51160 94330 57270 36575 95919 53092 18611 73819 32611
79310 51185 48074 46237 99627 49567 35188 57527 24891 22793 81830
11949 12983 36733 62440 65664 30860 21394 94639 52247 37190 70217
98609 43702 77053 92171 76293 17675 23846 74818 46766 94051 32000
56812 71452 63560 82778 57713 42757 78960 91736 37178 72146 84409
01224 95343 01465 49585 37105 07922 79689 25892 35420 19956 11212
90219 60864 03441 81598 13629 77477 13099 60518 70721 13499 99998
37297 80499 51059 73173 28160 96318 59502 44594 55346 90830 26425
22308 25334 46850 35261 93118 81710 10003 13783 87528 86587 53320
83814 20617 17766 91473 03598 25349 04287 55468 73115 95628 63882
35378 75937 51957 78185 77805 32171 22680 66130 01927 87661 11959
09216 42019 89380 95257 20106 54858 63278 86593 61533 81827 96823
03019 52035 30185 29689 95773 62259 94138 91249 72177 52834 79131
51557 48572 42454 15069 59508 29533 11686 17278 55889 07509 83817
54637 46493 93192 55060 40092 77016 71139 00984 88240 12858 36160
35637 07660 10471 01819 42955 59619 89467 67837 44944 82553 79774
72684 71040 47534 64620 80466 84259 06949 12933 13677 02898 91521
04752 16205 69660 24058 03815 01935 11253 38243 00355 87640 24749
64732 63914 19927 26042 69922 79678 23547 81636 00934 17216 41219
92458 63150 30286 18297 45557 06749 83850 54945 88586 92699 56909
27210 79750 93029 55321 16534 49872 02755 96023 64806 65499 11988
18347 97753 56636 98074 26542 52786 25518 18417 57467 28909 77772
79380 00816 47060 01614 52491 92173 21721 47723 50141 44197 35685
48161 36115 73525 52133 47574 18494 68438 52332 39073 94143 33454
77624 16862 51898 35694 85562 09921 92221 84272 55025 42568 87671
79049 46016 53466 80498 86272 32791 78608 57843 83827 96797 66814
54100 95388 37863 60950 68006 42251 25205 11739 29848 96084 12848
86269 45604 24196 52850 22210 66118 63067 44278 62203 91949 45047
12371 37869 60956 36437 19172 87467 76465 75739 62413 89086 58326
45995 81339 04780 27590 09946 57640 78951 26946 83983 52595 70982
58226 20522 48940 77267 19478 26848 26014 76990 90264 01363 94437
45530 50682 03496 25245 17493 99651 43142 98091 90659 25093 72216
96461 51570 98583 87410 59788 59597 72975 49893 01617 53928 46813
82686 83868 94277 41559 91855 92524 59539 59431 04997 25246 80845
98727 36446 95848 65383 67362 22626 09912 46080 51243 88439 04512
44136 54976 27807 97715 69143 59977 00129 61608 94416 94868 55584
84063 53422 07222 58284 88648 15845 60285 06016 84273 94522 67467
67889 52521 38522 54995 46667 27823 98645 65961 16354 88623 05774
56498 03559 36345 68174 32411 25150 76069 47945 10965 96094 02522
88797 10893 14566 91368 67228 74894 05601 01503 30861 79286 80920

```
87476 09178 24938 58900 97149 09675 98526 13655 49781 89312 97848
21682 99894 87226 58804 85756 40142 70477 55513 23796 41451 52374
62343 64542 85844 47952 65867 82105 11413 54735 73952 31134 27166
10213 59695 36231 44295 24849 37187 11014 57654 03590 27993 44037
42007 31057 85390 62198 38744 78084 78489 68332 14457 13868 75194
35064 30218 45319 10484 81005 37061 46806 74919 27819 11979 39952
06141 96634 28754 44064 37451 23718 19217 99983 91015 91956 18146
75142 69123 97489 40907 18649 42319 61567 94520 80951 46550 22523
16038 81930 14209 37621 37855 95663 89377 87083 03906 97920 77346
72218 25625 99661 50142 15030 68038 44773 45492 02605 41466 59252
01497 44285 07325 18666 00213 24340 88190 71048 63317 34649 65145
39057 96268 56100 55081 06658 79699 81635 74736 38405 25714 59102
89706 41401 10971 20628 04390 39759 51567 71577 00420 33786 99360
07230 55876 31763 59421 87312 51471 20532 92819 18261 86125 86732
15791 98414 84882 91644 70609 57527 06957 22091 75671 16722 91098
16909 15280 17350 67127 48583 22287 18352 09353 96572 51210 83579
15136 98820 91444 21006 75103 34671 10314 12671 11369 90865 85163
98315 01970 16515 11685 17143 76576 18351 55650 88490 99898 59982
38734 55283 31635 50764 79185 35893 22618 54896 32132 93308 98570
64204 67525 90709 15481 41654 98594 61637 18027 09819 94309 92448
89575 71282 89059 23233 26097 29971 20844 33573 26548 93823 91193
25974 63667 30583 60414 28138 83032 03824 90375 89852 43744 17029
13276 56180 93773 44403 07074 69211 20191 30203 30380 19762 11011
00449 29321 51608 42444 85963 76698 38952 28684 78312 35526 58213
14495 76857 26243 34418 93039 68642 62434 10773 22697 80280 73189
15441 10104 46823 25271 62010 52652 27211
```

6.5 La formule de Bailey-Borwein-Plouffe (BBP)

6.5.1 L'historique

En 1995, Plouffe (avec Bailey et Borwein) a découvert la formule de Bailey-Borwein-Plouffe (BBP) qui permet de calculer le n-ième bit de π sans avoir à calculer d'autres bits. Un an plus tard, il publie un nouvel article sur la formule, permettant de déterminer le n-ième chiffre en base 10 de π, mais le temps de calcul, bien que relativement court, n'est pas linéaire.

Il est également le co-auteur de l'Encyclopédie en ligne des suites de nombres entiers.

L'Inverseur de Plouffe est une page web qui contient plus de 200 millions de constantes mathématiques. Un répertoire est accessible et contient plus de 3,93 milliards de constantes à une précision de 64 chiffres décimaux au 21 juillet 2009.

Pour l'anecdote, Simon Plouffe a détenu en 1977 le record Guinness de mémorisation des décimales de π, avec 4 096 décimales. Il en avait mémorisé 4 400, mais en a récité seulement 4 096 parce que $4096 = 2^{12}$).

La formule de Plouffe ou formule BBP est

$$\pi = \sum_{n=0}^{+\infty} \frac{1}{16^k}(\frac{4}{8k+1} - \frac{2}{8k+4} - \frac{1}{8k+5} - \frac{1}{8k+6})$$

6.5.2 Démonstration de la formule de Plouffe en devoir (Université de Lorraine)

Le but de cet exercice est de montrer la formule de Plouffe

$$\pi = \sum_{k=0}^{+\infty} \frac{1}{16^k}(\frac{4}{8k+1} - \frac{2}{8k+4} - \frac{1}{8k+5} - \frac{1}{8k+6})$$

1. Montrer que pour tout a rèel avec $0 < a < 1$, pour tous $n, p \geq 1$ on a
$$\int_0^a \frac{x^{n-1}}{1-x^p} = \sum_{k=0}^{+\infty} \frac{a^{kp+n}}{kp+n}$$
2. En déduire que
$$\sum_{k=0}^{+\infty} \frac{1}{16^k}(\frac{4}{8k+1} - \frac{2}{8k+4} - \frac{1}{8k+5} - \frac{1}{8k+6}) = 16 \int_0^1 \frac{4-2x^3-x^4-x^5}{16-x^8}dx$$
3. Simplifier la fraction $\dfrac{4-2x^3-x^4-x^5}{16-x^8}$ (on pourra remarquer que le polynôme $4-2x^3-x^4-x^5$ a une racine simple), puis décomposer la fraction obtenue en éléments simples dans $\mathbb{Q}[X]$.
4. Conclure

Solution avec l'aide de `Xcas` On utilise ici `Xcas` en mode réel pour vérifier ou pour faire des calculs.

1. Pour $|x| < 1$ et $p > 1$, on a $|x^p| < 1$. On a donc le développement en série entière :
$$\frac{x^{n-1}}{1-x^p} = x^{n-1}\sum_{k=0}^{+\infty} x^{kp} = \sum_{k=0}^{+\infty} x^{kp+n-1}$$
Avec `Xcas`, on vérifie et on tape :
`assume(p,integer);sum(x^(k*p),k=0..inf)`
On obtient : `x^(-p)/(x^(-p)-1)`
A l'intérieur du rayon de convergence, on peut intégrer terme à terme et comme $\int_0^a x^{kp+n-1}dx = \frac{a^{kp+n}}{kp+n}$, on obtient :
$$\int_0^a \frac{x^{n-1}}{1-x^p}dx = \sum_{k=0}^{+\infty} \frac{a^{kp+n}}{kp+n}$$
Avec `Xcas`, on vérifie et on tape :
`int(x^(k*p+n-1),x=0..a)`
On obtient : `a^(k*p+n)/(k*p+n)`
2. Remarquons tout d'abord que :
$$\int_0^1 \frac{x^m}{16-x^8}dx = \int_0^1 \frac{x^m}{16(1-(\frac{x}{\sqrt{2}})^8}dx = \int_0^{\frac{\sqrt{2}}{2}} \frac{1}{16}\frac{u^m * \sqrt{2}^{m+1}}{1-u^8}du$$
Avec `Xcas`, on tape pour faire le changement de variable $x = u\sqrt{2}$ dans l'intégrale :

`subst('int(x^m/(16-x^8),x,0,1)',x=sqrt(2)*u)`
On obtient :
```
int((sqrt(2)*u)^m*1/(16-(sqrt(2)*u)^8)*sqrt(2),
            u,0,(sqrt(2))/2)
```
Puis, on sélectionne `sqrt(2)*u)^m*1/(16-(sqrt(2)*u)^8)*sqrt(2)`,
on appelle `simplify` et on obtient :
`int(-(sqrt(2)*u)^m*sqrt(2)/(16*u^8-16), u,0,(sqrt(2))/2)`
D'aprés ce qui précéde (pour $p = 8$, $m = n - 1$ et $a = \frac{\sqrt{2}}{2}$) et puisque :
$\sqrt{2}^{m+1} * a^{8k+m+1} = \dfrac{1}{\sqrt{2}^{8*k}} = \dfrac{1}{16^*k}$, on a :

$$\int_0^1 \frac{x^m}{16-x^8}dx = \frac{1}{16}\sum_{k=0}^{+\infty} \frac{1}{16^k*(8k+m+1)} = \frac{1}{16}\sum_{k=0}^{+\infty} \frac{1}{16^k*(8k+m+1)}$$

Donc :

$$16\int_0^1 \frac{4-2x^3-x^4-x^5}{16-x^8}dx = \sum_{k=0}^{+\infty}\frac{1}{16^k}\left(\frac{4}{8k+1} - \frac{2}{8k+4} - \frac{2}{8k+5} - \frac{1}{8k+6}\right)$$

3. **Factorisation de** $4 - 2x^3 - x^4 - x^5$
1 est racine évidente de $4 - 2x^3 - x^4 - x^5$.
On tape :
`factor(4-2x^3-x^4-x^5)`
On obtient :
`-(x-1)*(x^2+2)*(x^2+2*x+2)`
On a donc la factorisation

$$x^5 + x^4 + 2x^3 - 4 = (x-1)(x^2+2)(x^2+2x+2)$$

Factorisation de $16 - x^8$
On tape :
`factor(16-x^8)`
On obtient :
`-(x^2-2)*(x^2+2)*(x^2-2*x+2)*(x^2+2*x+2)`
Simplification de $\dfrac{4-2x^3-x^4-x^5}{16-x^8}$
On tape :
`normal((4-2x^3-x^4-x^5)/(16-x^8))`
On obtient :
`(x-1)/(x^4-2*x^3+4*x-4)`
Décomposition en éléments simples de $\dfrac{4-2x^3-x^4-x^5}{16-x^8}$
On tape :
`partfrac((x-1)/(x^4-2*x^3+4*x-4))`
On obtient :
`x/((x^2-2)*4)+(-x+2)/((x^2-2*x+2)*4)`
Ou bien , on tape directement :
`partfrac((4-2x^3-x^4-x^5)/(16-x^8))`
On obtient :
`x/((x^2-2)*4)+(-x+2)/((x^2-2*x+2)*4)`

4. **Conclusion**
   On a donc :
   $16\int_0^1 \frac{4-2x^3-x^4-x^5}{16-x^8}dx = 4\int_0^1 \frac{x}{x^2-2}dx + 4\int_0^1 \frac{-x+2}{x^2-2*x+2}dx =$
   $2\ln(|1^2-2|)-2\ln(|0^2-2|)-2\int_0^1 \frac{2x-2}{x^2-2*x+2}dx + 4\int_0^1 \frac{1}{x-1)^2+1}dx =$
   $-2\ln(2)+2\ln(2)+4*\text{atan}(0)-4*\text{atan}(-1)=\pi$
   On tape :
   `normal(16*int((4-2x^3-x^4-x^5)/(16-x^8),x=0..1))`
   On obtient :
   `pi`
   Donc :
   $$\pi = \sum_{n=0}^{+\infty} \frac{1}{16^k}\left(\frac{4}{8k+1}-\frac{2}{8k+4}-\frac{1}{8k+5}-\frac{1}{8k+6}\right)$$
   Avec `Xcas`, en mode réel, on peut taper directement :
   ```
   simplify(sum(1/16^k*(4/(8*k+1)-2/(8*k+4)-1/(8*k+5)-
   1/(8*k+6)),k=0..inf))
   ```
   On obtient :
   `pi`

6.5.3 Le n-ième chiffre après la virgule de π en base 16 avec la formule BBP

La formule de Bailey-Borwein-Plouffe (BBP) permet de calculer le n-ième chiffre après la virgule de π en base 16 sans avoir à calculer d'autres chiffres.
On remarque que si $n \geq 1$, le $n+1$-ième chiffre après la virgule de π en base 16 est aussi le 1-ier chiffre après la virgule de $16^n\pi$.
On cherche donc 1-ier chiffre après la virgule de $16^n\pi$ en base 16 et on a :

$$16^n\pi = \sum_{n=0}^{+\infty} 16^{n-k}\left(\frac{4}{8k+1}-\frac{2}{8k+4}-\frac{1}{8k+5}-\frac{1}{8k+6}\right)$$

Exemple d'écriture en base 16 d'une approximation de π
Avec `Xcas`, on tape :
```
Digits:=25
a:=evalf(pi)
```
On obtient :
`3.141592653589793238462643`
On calcule 16^ka pour $k = 0..21$, on tape :
`L:=floor(16^k*a)$(k=0..21)`
Puis on convertit le résultat obtenu en base 16 :
`LB:=revlist(convert(L[k],base,16))$(k=0..21))`
On obtient :
```
[3],[3,2],[3,2,4],[3,2,4,3],[3,2,4,3,15],[3,2,4,3,15,6],
[3,2,4,3,15,6,10],[3,2,4,3,15,6,10,8],[3,2,4,3,15,6,10,8,8],
[3,2,4,3,15,6,10,8,8,8],[3,2,4,3,15,6,10,8,8,8,5]...
```
On tape :

`L1:=LB[21];`
On obtient :
`[3,2,4,3,15,6,10,8,8,8,5,10,3,0,8,13,3,1,3,1,9,8]`
Le k-ième chiffre après la virgule de π en base 16 est `L1[k]`
On vérifie, on tape :
`evalf(sum(L1[k]/16^k,k=0..10),evalf(pi))`
On obtient avec `Digits:=21` :
`3.14159265358979323846264 34,3.14159265358979323846264 34`
On utilise maintenant la formule BBP pour calculer le k-ième chiffre après la virgule de π en base 16.
Posons :

$$S_n(a) = \sum_{k=0}^{\infty} \frac{16^{n-k}}{8k+a}$$

Le calcul des premiers chiffres de $S_n(a)$ permettra d'obtenir ceux de $16^n\pi$, par la relation :
$16^n\pi = 4S_n(1) - 2S_n(4) - S_n(5) - S_n(6)$
Découpons la somme $S_n(a)$ en deux sommes (l'une lorsque k varie entre 0 et $n-1$ et l'autre lorsque k varie entre n et $+\infty$:

$$S_n(a) = \sum_{k=0}^{\infty} \frac{16^{n-k}}{8k+a} = \sum_{k=0}^{n-1} \frac{16^{n-k}}{8k+a} + \sum_{k=n}^{\infty} \frac{16^{n-k}}{8k+a} = A_n(a) + B_n(a)$$

Calculons les premiers chiffres de $A_n(a)$ et $B_n(a)$.

Les premiers chiffres après la virgule de $4A_n(1)-2A_n(4)-A_n(5)-A_n(6)$
Il faut trouver la partie fractionnaire de $\frac{16^{n-k}}{8k+a}$.
La partie entière de $\frac{16^{n-k}}{8k+a}$ est tres grande et on va utiliser l'arithmétique modulaire pour la soustraire à $\frac{16^{n-k}}{8k+a}$.
On utilise, pour cela, la division euclidienne de 16^{n-k} par $8k+a$:

$$16^{n-k} = q(8k+a) + r \text{ avec } q \in \mathbb{Z} \text{ et } r < 8k+a, r \in \mathbb{Z}$$

c'est à dire $16^{n-k} = r \bmod (8k+a)$.
Donc $\frac{16^{n-k}}{8k+a} = q + \frac{r}{8k+a}$.
Comme $q \in \mathbb{Z}$ et $\frac{r}{8k+a} < 1$, la partie fractionnaire de $\frac{16^{n-k}}{8k+a}$ est $\frac{r}{8k+a}$ avec $16^{n-k} = r \bmod 8k+a$.
On veut avoir les premiers chiffres après la virgule de $\frac{r}{8k+a}$ si $16^{n-k} = r \bmod 8k+a$.
Il faut donc calculer la partie fractionnaire de la suite SA (le nom de cette suite sur deux lettres fait référence à une variable d'une commande Xcas plus bas) définie par :

$$SA_n = 4\tilde{A}_n(1) - 2\tilde{A}_n(4) - \tilde{A}_n(5) - \tilde{A}_n(6)$$

où :

$$\tilde{A}_n(a) = \sum_{k=0}^{n-1} \frac{(16^{n-k} \bmod (8k+a))}{8k+a}$$

Attention
Si `frac`$(SA_n < 0)$ alors la partie fractionnaire de :

$4A_n(1) - 2A_n(4) - A_n(5) - A_n(6)$ vaut 1+`frac`(SA_n) et
Si `frac`$(SA_n > 0)$ alors la partie fractionnaire de :
$4A_n(1) - 2A_n(4) - A_n(5) - A_n(6)$ vaut `frac`(SA_n)

Exemple

Si $n = 10$ et $a = 1$, pour obtenir le 11-ième chiffre apres la virgule de π en base 16, on doit calculer la partie fractionnaire de :
$4\tilde{A}_{10}(1) - 2\tilde{A}_{10}(4) - \tilde{A}_{10}(5) - \tilde{A}_{10}(6)$ avec
$\tilde{A}_{10}(a) = \sum_{k=0}^{9} \frac{(16^{10-k} \bmod (8k+a))}{8k+a}$.
On tape pour avoir $4\tilde{A}_{10}(1) - 2\tilde{A}_{10}(4) - \tilde{A}_{10}(5) - \tilde{A}_{10}(6)$:

```
SA:=evalf(sum(4*powmod(16,10-k,8k+1)/(8k+1)-
2*powmod(16,10-k,8k+4)/(8k+4)-powmod(16,10-k,8k+5)/(8k+5)-
powmod(16,10-k,8k+6)/(8*k+6),k=0..9))
```

On obtient `-2.365446858202734074099452 4`
Donc la partie fractionnaire de $4A_{10}(1) - 2A_{10}(4) - A_{10}(5) - A_{10}(6)$ est donc :
`3+SA` soit `3-2.36544685820273407=0.6345531417973`

Les premiers chiffres après la virgule de $4B_n(1) - 2B_n(4) - B_n(5) - B_n(6)$
Posons : $b_k = \frac{16^{n-k}}{8k+a}$
Le premier terme de la somme $B_n(a)$ est :
$b_n = \frac{1}{8n+a}$ et donc $b_n < 1$
Calculons $\frac{b_k}{b_{k+1}}$:
$\frac{b_k}{b_{k+1}} = \frac{16(8k+8+a)}{8k+a} = 16(1 + \frac{8}{8k+a}) > 16$ Donc pour obtenir $B_n(a)$ avec une précision de p chiffres après la virgule, on va calculer les $p + 10$ premiers termes de la somme pour tenir compte des retenues éventuelles .
Il suffit donc de calculer :
$\tilde{B}_n(a) = \sum_{k=n}^{n+p+10} \frac{16^{n-k}}{8k+a}$
Exemple (suite)
Si $n = 10$, $p = 4$ et $a = 1$, on a :
$\tilde{B}_{10}(1) = \sum_{k=10}^{10+4+10} \frac{16^{10-k}}{8k+1}$
On tape pour avoir $4\tilde{B}_{10}(1) - 2\tilde{B}_{10}(4) - \tilde{B}_{10}(5) - \tilde{B}_{10}(6)$:

```
SB:= evalf(sum(4*16^(10-k)/(8k+1)-2*16^(10-k)/(8k+4)-
16^(10-k)/(8k+5)-16^(10-k)/(8k+6),k=10..24))
```

On obtient `0.2300259542292191541194 4196e-2`

Conclusion Au final, pour obtenir les n premiers chiffres de π en base 16, il faut calculer les premiers chiffres après la virgule en base 16 de :
$\pi_n = 4\tilde{S}_n(1) - 2\tilde{S}_n(4) - \tilde{S}_n(5) - \tilde{S}_N(6)$
avec $\tilde{S}_n(a) = \displaystyle\sum_{k=0}^{n-1} \frac{16^{n-k}[8k+a]}{8k+a} + \sum_{k=n}^{n+p+10} \frac{16^{n-k}}{8k+a}$
Fin de l'exemple
On tape :

```
3+SA+SB
```

On obtient `0.6368534013395581174417 4205`
On tape pour avoir le 11-ième chiffre après la virgule en base 16 de π i.e. le premier chiffre après la virgule en base 16 de $3 + SA + SB$, on tape :

```
floor(16*(3+SA+SB)),L1[11]
```

On obtient `10,10`
Puisque $p = 4$, on tape pour avoir les quatre premiers chiffres après la virgule en

base 16 de `revlist(convert(floor(16^4*(3+SA+SB)),base,16))`
On obtient `[10,3,0,8]`
On vérifie en tapant `[L1[k]$(k=11..14)]`
On obtient bien :
`[10,3,0,8]`
Un autre exemple pour avoir de la 13-ième à la 16-ième décimale en base 16 de π
On tape ($n = 12$) :

```
SA2:=evalf(sum(4*powmod(16,12-k, 8k+1)/(8*k+1)-
2*powmod(16,12-k,8k+4)/(8k+4)- powmod(16,12-k,8k+5)/(8k+5)-
powmod(16,12-k,8k+6)/(8*k+6),k=0..11))
```

On obtient :
`-0.119671481184960585697030054e2`
On tape ($n = 12$ et $p = 4$) :

```
SB2:=evalf(sum(4*16^(12-k)/(8k+1)- 2*16^(12-k)/(8k+4)-
16^(12-k)/(8k+5)- 16^(12-k)/(8k+6),k=12..26))
```

On obtient :
`0.161886142293663487457435654e-2`
On tape :
`floor(16^k*(12+SA2+SB2))$(k=1..4)`
On obtient :
`0,8,141,2259`
On tape :

```
revlist(convert(0,base,16)),revlist(convert(8,base,16)),
revlist(convert(141,base,16)),revlist(convert(2259,base,16))
```

On obtient :
`[8,13],[8,13,3]`
Attention il ne faut pas oublier les zéros éventuels qui ne sont pas marqués ! ! ! car on doit renvoyer une liste de longueur 1 pour la 12-ième... et une liste de longueur 4 pour la 15-ième décimale : `[0],[0,8],[0,8,13],[0,8,13,3]`
Donc la 12-ième décimale est 0 la 14-ième est 8 et les suivantes sont 13 et 3
On vérifie :
`L1[k]$(k=13..16)`
On obtient bien :
`0,8,13,3` **Exercice**
Trouver la 10000-ième décimale de π en base 16.

On pose $n = 10^4 - 1$ et $p = 4$ et on tape :

```
n:=10000-1
SA3:=evalf(sum((4*powmod(16,n-k, 8k+1)/(8*k+1)-
2*powmod(16,n-k, 8k+4)/(8k+4)- powmod(16,n-k,8k+5)/(8k+5)-
powmod(16,n-k,8k+6)/(8*k+6)),k,0,(n-1)))
```

On obtient :
`0.634088830810459669232742620e2`
On tape :

```
SB3:=evalf(sum(4*16^(n-k)/(8k+1)-
2*16^(n-k)/(8k+4)-16^(n-k)/(8k+5)- 16^(n-k)/(8k+6),k=n..n+14))
```

On obtient :
`0.250028128086193875462763370e-8`

On tape :
`revlist(convert(floor(16^4*(frac(SA3)+SB3)),base,16))`
On obtient la 10000-ième, 10001-ième, 10002-ième,10003 ième décimale de π en base 16 :
`[6,8,10,12]`

6.5.4 Le programme `Xcas`

À l'aide de la formule BBP, on écrit un programme pour trouver, le n-ième,... le $n+p-1$-ième chiffre après la virgule, en base 16, de π.
On tape :

```
pichiffre16(n,p):={
local SA,SB,L,s,k;
si n<0 alors retourne "n doit etre >=0" fsi;
n:=n-1;
SA:=sum((4*powmod(16,n-k, 8k+1)/(8*k+1)-
     2*powmod(16,n-k, 8k+4)/(8k+4)-
     powmod(16,n-k,8k+5)/(8k+5)-
     powmod(16,n-k,8k+6)/(8*k+6)),k,0,(n-1));
SA:=frac(SA);
si SA<0 alors SA:=1+SA fsi;
SB:=sum(4*16^(n-k)/(8k+1)-2*16^(n-k)/(8k+4)
    -16^(n-k)/(8k+5)-16^(n-k)/(8k+6),k=n..n+p+10);
L:=revlist(convert(floor(16^p*frac(SA+SB)),base,16));
s:=size(L);
pour k de 0 jusque p-s-1 faire L:=prepend(L,0); fpour;
retourne L;
}:;
```

On tape :
`pichiffre16(13,5)`
On obtient : `[0,8,13,3,1]`
On tape pour avoir l'ècriture en base 16 d'une approximation de π :
`pichiffre16(0,25)`
On obtient : `[3,2,4,3,15,6,10,8,8,8,5,10,3,0,8,13,3,1,3,1,9,8,10,2,14]]`
On tape :
`pichiffre16(10000,4)`
On obtient : `[6,8,10,12]`

6.5.5 Formule d'Adamchick et Wagon

$$\pi = \sum_{k=0}^{\infty} \frac{(-1)^k}{4^k} \left(\frac{2}{4k+1} + \frac{2}{4k+2} + \frac{1}{4k+3} \right)$$

Avec `Xcas`, on tape :
`simplify(sum((-1)^k/4^k*(2/(4k+1)+2/(4k+2)+1/(4k+3)),k,0,inf))`
On obtient : `1/2*pi+atan(1/2)+atan(2)`
On tape :

`simplify(1/2*pi+atan(1/2)+atan(2))`
On obtient : `pi`

6.5.6 Formule de Plouffe généralisée

Notons $S_n = \sum_{k=0}^{\infty} \frac{1}{16^k(8k+n)}$.
La formule de Plouffe généralisée est :

(0) $\forall r \in \mathbb{C} \quad \pi = (4{+}8r)S_1{-}8rS_2{-}4rS_3{-}(2{+}8r)S_4{-}(1{+}2r)S_5{-}(1{+}2r)S_6{+}rS_7$

Avec `Xcas`, on tape :

```
simplify(sum(1/16^k*((4+8r)/(8k+1)-8r/(8k+2)-4r/(8k+3)-
(2+8r)/(8k+4)-(1+2r)/(8k+5)-(1+2r)/(8k+6)+r/(8k+7)),k,0,inf)
```

On obtient : `pi`
Démonstration
Pour $r = 0$ on retrouve la formule BBP ;
Pour $r = -1/4$ on retrouve la formule d'Adamchick et Wagon sous une forme plus détaillée.
Posons :
$\alpha = 1 - i = \sqrt{2}e^{-i\pi/4}$ et calculons de deux façons l'intégrale :

$$I = \int_0^1 \frac{dy}{\alpha - y}$$

. Elle est reliée aux S_n car :

$$I = \frac{1}{\alpha}\int_0^1 \frac{dy}{1 - y/\alpha} = \frac{1}{\alpha}\int_0^1 \sum_{m=0}^{\infty} \frac{y^m}{\alpha^m} dy = \sum_{m=0}^{\infty} \frac{e^{i(m+1)\pi/4}}{(m+1)\sqrt{2}^{m+1}} =$$

$$\frac{1+i}{2}S_1 + \frac{i}{2}S_2 + \frac{-1+i}{4}S_3 - \frac{1}{4}S_4 - \frac{1+i}{8}S_5 - \frac{i}{8}S_6 + \frac{1-i}{16}S_7 + \frac{1}{16}S_8,$$

Elle calculable par des méthodes élémentaires (en calculant séparément sa partie réelle et sa partie imaginaire ou avec un logarithme complexe), ou avec `Xcas` en mode complexe :

$$I = -\left[\ln(\alpha - y)\right]_0^1 = \ln\left(\frac{\alpha}{\alpha - 1}\right) = \ln(1+i) = \ln(\sqrt{2}e^{i\pi/4}) = \frac{\ln 2}{2} + i\frac{\pi}{4}$$

Ou on tape en mode complexe :
`normal(int(1/(1-i-y),y,0,1))`
On obtient : `i/4*pi+1/2*ln(2)`
L'égalité entre ces deux expressions de I équivaut à :

(1) $\pi = 4\,Im(I) = 2S_1 + 2S_2 + S_3 - \frac{1}{2}S_5 - \frac{1}{2}S_6 - \frac{1}{4}S_7$,

(2) $\ln(2) = 2\,Re(I) = S_1 - \frac{1}{2}S_3 - \frac{1}{2}S_4 - \frac{1}{4}S_5 + \frac{1}{8}S_7 + \frac{1}{8}S_8$

Mais $\ln(2)$ peut s'exprimer en fonction des S_n :

(3) $\int_0^1 \frac{2y}{2-y^2}dy = [-\ln(2{-}y^2)]_0^1 = \ln(2)\int_0^1 \frac{y}{1-y^2/2}dy = \sum_{k\geq 0} \frac{1}{(2k+2)2^k}$
$= S_2 + \frac{1}{2}S_4 + \frac{1}{4}S_6 + \frac{1}{8}S_8$

On soustrait (2) et (3) et on a une relation entre les S_n :

(4) $0 = 2S_1 - 2S_2 - S_3 - 2S_4 - \frac{1}{2}S_5 - \frac{1}{2}S_6 + \frac{1}{4}S_7$

En multipliant (4) par $1 + 4r$ et en ajoutant ce produit à (1), on obtient l'égalité (0).

Chapitre 7

Les entiers de Gauss et l'algorithme de Todd

Merci à Bernard Ycart de nous avoir envoyé le fichier de l'article de Raymond Seroul sur les formules à la machin et l'algorithme de Todd, paru en 1986 à l'IREM de Strasbourg.

7.1 $\mathbb{Z}[i]$ ou les entiers de Gauss $\mathbb{Z}[i]$

7.1.1 La division euclidienne et le pgcd dans $\mathbb{Z}[i]$

On rappelle que $\mathbb{Z}[i] = \{m + in,\ (m, n) \in \mathbb{Z}^2\}$.
$\mathbb{Z}[i]$ est un anneau intègre euclidien.
Soient $a \in \mathbb{Z}[i]$ et $b \in \mathbb{Z}[i] - \{0\}$, alors on dit que le quotient entier q de a par b est l'affixe du (ou des) point(s) le plus proche pour le module du point d'affixe a/b et alors le reste de la division euclidienne est $r = a - bq$.
On choisit q pour que bq soit le plus proche possible de a et on peut montrer que l'on peut choisir $r = a - bq$ tel que $|r|^2 \leq |b|^2/2$.
Le pgcd dans $\mathbb{Z}[i]$ se calcule comme dans $\mathbb{N}$ par l'algorithme d'Euclide.

7.1.2 Les fonctions iquo, irem iquorem et gcd de `Xcas`

Si a et b sont des entiers ou des entiers de Gauss :
`iquo(a,b)` renvoie le quotient q de la division euclidienne de a par b et
`irem(a,b)` renvoie le reste r de la division euclidienne de a par b.
q et r vérifient :
si a et b sont entiers $a = bq + r$ avec $0 \leq r < b$
si a et b sont des entiers de Gauss $a = bq + r$ avec $|r|^2) \leq \frac{|b|^2}{2}$.
`iquorem(a,b)` renvoie la liste $[q, r]$ du quotient et du reste de la division euclidienne de a par b .
`gcd(a,b)` renvoie le pgcd de a et b
Par exemple :
si $a = -6 + 17i$ et si $b = 7 + i$ on tape :
`iquo(-6+17*i,7+i)`
et on obtient : `-1+3*i`
`irem(-6+17*i,7+i)`

et on obtient : `4-3*i`
`iquorem(-6+17*i,7+i)`
et on obtient : `[-1+3*i,4-3*i]`
Donc la division euclidienne de a par b a pour quotient $-1+3i$ et pour reste $4-3i$
.
On tape : `iquorem(7+i,4-3*i)`
et on obtient : `[1+i,0]`
Donc $7+i$ est un multiple de $4-3i$ `gcd(7+i,-6+17*i)`
et on obtient : `4-3*i`

7.1.3 Exercice

— Soit $q \in \mathbb{Z}[i]$ tel que bq soit le plus proche possible de a et $r = a - bq$. Montrer que l'on peut choisir r tel que $|r|^2 \leq |b|^2/2$.

— Montrer que $|q - a/b|^2 \leq 1/2$. En déduire que $|a - bq|^2 \leq |b|^2/2$ et que l'algorithme d'Euclide se termine lorsqu'on prend q comme quotient euclidien.
— Écrire un programme qui calcule le pgcd de 2 nombres de $\mathbb{Z}[i]$. On normalisera le résultat (en multipliant le résultat par 1,-1,i ou -i) pour que le pgcd soit un nombre de partie rèelle strictement positive et de partie imaginaire positive ou nulle.

On tape :

```
quotient(a,b):={
local q1,q2,c;
c:=normal(a/b);
q1:=re(c);
q2:=im(c);
return round(q1)+i*round(q2);
}
:;
reste(a,b):={
local q;
q:=quotient(a,b);

return a-b*q;
}
:;

pgcdzi(a,b):={
local q,r;
tantque b!=0 faire
  q:=quotient(a,b);
  r:=a-b*q;
  a:=b;
  b:=r;
ftantque;
```

```
//on normalise
si re(a)<0 et im(a)<=0 alors retourne -a;fsi;
si im(a)<0 alors retourne i*a;fsi;
si re(a)<=0 alors retourne -i*a;fsi;
retourne a;
}:;
```

On tape :
`pgcdzi(3+i,3-i)`
On obtient :
`1+i`
On tape :
`pgcdzi(7+i,-6+17*i`
On obtient :
`3+4*i`

7.1.4 La fonction pa2b2 de `Xcas`

Théorème
Si p est un nombre premier de $\mathbb{N}$ congru à 1 modulo 4, alors il existe deux entiers a et b tel que $p = a^2 + b^2$.
On tape :
`pa2b2(89)`
On obtient :
`[5,8]`
On tape :
`pa2b2(317)`
On obtient :
`[11,14]`
Idée de la preuve du théorème
Dans ce qui suit, p est un nombre premier de $\mathbb{N}$ congru à 1 modulo 4.

— **Lemme**
  -1 admet une racine carrée modulo p.
  Preuve du lemme
  On rappelle le théorème de **Wilson** qui dit que si $p \geq 3$ est premier alors $(p-1)! = -1 \bmod p$.
  (En effet $\mathbb{Z}/p\mathbb{Z} - \{0\}$ est un group multiplicatif donc si $a = (p-1)!$ on a $a^2 = 1 \bmod p$ car on peut associer chaque facteur de a avec son inverse dans $\mathbb{Z}/p\mathbb{Z} - \{0\}$.Donc $a^2 - 1 = (a-1) * (a+1) = k * p$. Comme p est premier, p divise $a-1$ ou $a+1$.)
  Posons $p = 2q+1$. Puisque $p = 1 \bmod 4$ on a q est pair.
  On a :
  $q! = (q-p)(q-1-p)...(2-p)(1-p) \bmod p = (-1)^q(q+1)(q+2)...(2q-1)(2q) \bmod p$ donc :
  $q!^2 = (1 * 2 * ...q) * ((-1)^q(q+1)(q+2)...(2q-1)(2q)) \bmod p$ donc puisque q est pair :
  $q!^2 = (p-1)! \bmod p = -1 \bmod p$ On notera $rac = q! \bmod p$.
  Par exemple pour $p = 5$ $rac = 2$, pour p=pour $p = 17$ $rac = 13$.

— On pose $u = [1, rac]$ et $v = [0, p]$ et on considère le réseau engendré par u et v : $\Lambda = \mathbb{Z}u + \mathbb{Z}v$. On cherche une autre base du réseau dont un vecteur est de longueur minimale. Quitte à échanger u et v, on peut supposer que $v * v \leq u * u$ ($*$ est le produit scalaire).
On montre facilement que si on remplace (u, v) par $(u + mv, v)$ avec $m \in \mathbb{Z}$, le réseau engendré est identique.
On montre facilement que si $w = [w_1, w_2] \in \Lambda = \mathbb{Z}u + \mathbb{Z}v$ alors le produit scalaire $w * w$ est divisible par p (on a en effet $1 + rac^2 = 0 \bmod p$ donc : $w * w = w_1^2(1 + rac^2) + w_2^2(p^2) + 2w_1w_2p * rac = 0 \bmod p$).
On montre facilement que tous les parallélogrammes engendrés par une base u_1 et $v - 1$ du réseau ont une aire égale à p (car le produit vectoriel de u et v a comme longueur p et il est égal au produit vectoriel de $u + mv$ et v et est aussi égal au produit vectoriel de $v + mu$ et u).
On choisit $m \in \mathbb{Z}$ pour que la projection de $w = u + m * v$ sur v soit dans $] - v/2, v/2]$ (i.e.$-(v * v)/2 < w * v = (u + m * v) * v \leq (v * v)/2$ où $*$ est le produit scalaire ou encore $m = floor(1/2 - u * v/v * v) =$.
On a alors $w * w \leq u * u$.
En effet :
$v * v \leq u * u$ et
$|w * v| = (u + mv) * v \leq (v * v)/2$ (d'après le choix de m
Puisque $u * v$ n'est pas dans le segment $]] - v * v/2; v * v/2$ c'est que :
soit $u * v > v * v/2$ et alors $m < 0$
soit $u * v \leq -v * v/2$ et alors $m > 0$
On a :
$w*w = (u+mv)*w = u*w++m*(v*w) == u*u+m(u*v+w*v)$
si $m > 0$ alors $u * v + w * v < -v * v/2 + (v * v)/2 \leq 0$ et
si $m < 0$ alors $u * v + w * v > v * v/2 - (v * v)/2 \geq 0$ donc
dans les 2 cas $m(u * v + w * v) \leq 0$ i.e. $w * w \leq u * u$
On recommence l'opération jusqu'à ce que u soit dans la bande ($m = 0$).
On montre alors que la distance minimale entre 2 points du réseau est ègale à la norme du dernier v.
En effet lorsque $m = 0$ c'est que l'on est dans la situation suivante :
$v * v <= u * u$ et $|u * v| \leq v * v/2$ donc l'aire (égale à $p/2$) du triangle engendré par u et v est supérieure à l'aire du triangle équilatéral T de côté v. T a comme aire $A = (v * v) * \sqrt{3}/4$.



Donc on a $(v * v) * \sqrt{3}/2 \leq p$ puisque :
pour tous les nombres w du réseau, on a $w * w$ est divisible par p donc si $v = (a, b)$ on a $v * v = a^2 + b^2 < 2p$ et $v * v$ est un entier divisible par p donc $v * v = a^2 + b^2 = p$.
Remarque Lorsque l'algorithme s'arrête les vecteur u et v ont même norme et sont perpendiculaires.
En effet lorsque l'algorithme s'arrête on a :

1. $v * v = p$,
2. $|u * v| \leq v * v/2$ donc si α est l'angle formé par u et v on a : $\cos(\alpha)^2 \leq 1/4$ ou $\sin(\alpha)^2 \geq 3/4$,
3. $u * u$ est divisible par p ($u * u = kp$ avec $k \in \mathbb{N}^*$) et $u * u \geq v * v$,
4. l'aire A du parallélogramme engendré par u et v vaut p
5. $v * v = p$

Or $A^2 = \sin(\alpha)^2 (u * u)(v * v) = p^2 = \sin(\alpha)^2 (u * u)p$
Donc : $\sin(\alpha)^2 (u * u) = p = \sin(\alpha)^2 kp$
Donc : $k \sin(\alpha)^2 = 1 \geq 3k/4$ avec $k \in \mathbb{N}^*$
Ou encore $k \in \mathbb{N}$ et $k \leq 4/3$ donc $k = 1$ ce qui signifie que :
$u * u = p$ et $\sin(\alpha)^2 = 1$ donc $u * v = 0$

L'algorithme sur un exemple
On choisit $p = 13$, et prenons $rac = 5$ (puisque $5^2 = 25 = -1 \bmod 13$).
Dessinons quelques points du reseau en tapant :

```
reseau(p,rac):={
local L,j,k;
L:=NULL;
pour (j:=-2;j<=2;j++) {
  pour (k:=-3;k<=3;k++) {
    L:=L,point(k+i*(rac+k*p)+13*j);
  }
}
return affichage(L,epaisseur_point_2+1);
}:;
```



On commence avec $u = 1, rac$ et $v = 0, p$. on réduit u, v ce qui veut dire que l'on échange u et v (car $u * u < v * v$), puis on remplace v par $w = v - 2 * u, u$ afin que w soit dans la bande bleue.
On réduit w, u ce qui veut dire que l'on remplace w, u par u, w (car $u * u > w * w$) puis on remplace u par $t = u - w$ afin que t soit dans la bande verte. On s'arrête et t et w sont de norme minimale ($t * t = w * w = 2^2 + 3^2 = 13$).

L'algorithme

1. On écrit la fonction puissance rapide `puiss(a,k,p)` qui renvoie $a^k \bmod p$ (ou on utilise la fonction `powmod` de `Xcas`).
2. On écrit la fonction `sqrtmod(p)` qui renvoie une racine carrée de -1 modulo p lorsque p est congru à 1 modulo 4.
   Pour $p = 13$, `sqrtmod(13)` qui renvoie 8.
3. On écrit la fonction `reduire(u,v)` qui prend en entrée deux vecteurs u et v de $\mathbb{Z}^2$, qui échange u et v si $v * v > u * u$ et renvoie $m, u1, v1$ tels que $-(v*v)/2 < u+m*v \leq (v*v)/2$ (le produit étant le produit scalaire).
   Pour $p = 13$, `reduire([1,8],[0,13])` renvoie `-2,[-2,-3],[1,8]`, (car $[0, 13] - 2 * [1, 8] = [-2, -3]$)
   `reduire([-2,-3],[1,8])` renvoie `2,[-3,2],[-2,-3]` et
   `reduire([-3,2],[-2,-3])` renvoie `0,[-3,2],[-2,-3]`
   ou si on a pris $rac = 5$
   `reduire([1,5],[0,13])` renvoie `-2,[-2,-3],[1,5]` et
   `reduire([-2,-3],[1,5])` renvoie `-1,[-3,2],[-2,-3]` et
   `reduire([-3,2],[-2,-3])` renvoie `0,[-3,2],[-2,-3]`
4. On écrit la fonction `solpa2b2(p)` qui renvoie deux entiers a et b tels que $p = a^2 + b^2$ lorsque p est premier et congru à 1 modulo 4.
   On itère pour cela la fonction `reduire` avec comme arguments initiaux $u = [1, sqrtmod(p)]$ et $v = [0, p]$ et on s'arrête quand $m = 0$. On renvoie alors les coordonnées en valeurs absolues du dernier v. Pour $p = 13$, `solpa2b2(13)` renvoie $[3, 2]$.

Traduction de l'algorithme en langage `Xcas`

```
puiss(a,k,p):={
local pui;
pui:=1;
while(k>0) {
if (irem(k,2)!=0){
pui:=irem(a*pui,p);
k:=k-1;}
a:=irem(a*a,p);
k:=k/2;
}
return pui;
}:;
sqrtmod(p):={
local j,k,a,r;
if (!isprime(p)) {return p+"n'est pas premier"};
k:=iquo(p-1,4);
r:=irem(p-1,4);
if (r!=0) {return "erreur"};
a:=2;
j:=puiss(a,k,p);
while (j == 1 or j==p-1){
a:=a+1;
j:=puiss(a,k,p);
}
return j;
}:;
reduire(u,v):={
local w,uv,m,v2;
if (u*u<v*v){
w:=u;
u:=v;
v:=w;
}
uv:=u*v;
v2:=v*v;
m:=floor(1/2-uv/v2);
return m,u+m*v,v;
}:;
solpa2b2(p):={
local u,v,m,rac;
rac:=sqrtmod(p);
u:=[1,rac];
v:=[0,p];
m,u,v:=reduire(u,v);
while (m!=0) {
m,u,v:=reduire(u,v);
}
```

```
return abs(v);
}:;
```

On tape :
`solpa2b2(1009)`
On obtient :
`[15,28]`
On tape :
`pa2b2(1009)`
On obtient :
`[28,15]`
On tape :
`solpa2b2(1000033)`
On obtient :
`[408,913]`
On tape :
`pa2b2(1000033)`
On obtient :
`[913,408]`
On tape :
`solpa2b2(1000000009)`
On obtient :
`[3747,31400]`
On tape :
`pa2b2(1000000009)`
On obtient :
`[31400,3747]`
On tape :
`solpa2b2(1000000000061)`
On obtient :
`[848494,529205]`
On tape :
`pa2b2(1000000000061)`
On obtient :
`[529205,848494]`

7.2 Les nombres irréductibles de $\mathbb{Z}[i]$

7.2.1 Définitions et Théorème

Définition
On dit que $a \in \mathbb{Z}[i]$ est **inversible** si il existe $b \in \mathbb{Z}[i]$ tel que $ab = 1$.
On dit que $b \in \mathbb{Z}[i]$ est **associé** à $a \in \mathbb{Z}[i]$ si il existe $c \in \mathbb{Z}[i]$ inversible tel que $a = bc$.
Exercice
Déterminer les éléments inversibles de $\mathbb{Z}[i]$.
Déterminer les associé de $x + iy \in \mathbb{Z}[i]$.
Solution

Si $a = a_1 + ia_2 \in \mathbb{Z}[i]$ est inversible si il existe $b = b_1 + ib_2 \in \mathbb{Z}[i]$ tel que $ab = 1$.
Donc en égalant les modules :
$1 = (a_1^2 + a_2^2)(b_1^2 + b_2^2)$ comme $a_1^2 + a_2^2 \in \mathbb{N}$ et $b_1^2 + b_2^2 \in \mathbb{N}$ cela entraine que $a_1^2 + a_2^2 = 1$ c'est à dire $a \in \{1, i, -1, -i\}$ Les associés de $x + iy$ sont donc $\{x + iy, -x - iy, y + ix, y - ix\}$.

Définition

On dit que $a \in \mathbb{Z}[i]$ est **irréductible** si a est non inversible ou si a n'est le produit de 2 nombres appartenant à $\mathbb{Z}[i]$ que lorsque l'un de ces nombres est inversible.
Par exemple :
Les éléments inversibles de $\mathbb{Z}[i]$, $\{1, -1, i, -i\}$, ne sont pas irréductibles dans $\mathbb{Z}[i]$.
2 n'est pas irréductible dans $\mathbb{Z}[i]$ car $2 = (1 + i) * (1 - i)$.
$1 + 2i$ est irréductible car si $1 + 2i = (a_1 + ia_2)(b_1 + ib_2)$ alors en égalant le carré des modules on a :
$5 = (a_1^2 + a_2^2)(b_1^2 + b_2^2)$. Comme 5 est premier cela entraine que $a_1^2 + a_2^2 = 1$ (ou $b_1^2 +_2^2 = 1$) c'est à dire $a \in \{1, i, -1, -i\}$ (ou $b \in \{1, i, -1, -i\}$)
$(1 + i)$ est irreductible car si $1 + i = (a_1 + ia_2)(b_1 + ib_2)$ en égalant le carré des modules on obtient :
$2 = (a_1^2 + a_2^2)(b_1^2 + b_2^2)$ donc $(a_1^2 + a_2^2) = 1$ et $(b_1^2 + b_2^2) = 2$ (ou $(b_1^2 + b_2^2) = 1$ et $(a_1^2 + a_2^2) = 2$) ce qui veut dire que $(a_1 + ia_2)$ est inversible et $(b_1 + ib_2)$ est l'associé de $1 + i$ (ou $(b_1 + ib_2)$ est inversible et $(a_1 + ia_2)$ est l'associé de $1 + i$).

Théorème

Si dans $\mathbb{Z}[i]$ a irréductible divise bc alors a divise b ou a divise c.

Théorème

Si un entier de Gauss est irréductible alors l'un de ses associé est :

1. le nombre $1 + i$
2. $x + iy$ où $x^2 + y^2$ est un nombre premier de $\mathbb{Z}$ congru à 1 modulo 4
3. un nombre premier de $\mathbb{Z}$ congru à 3 modulo 4

Ainsi un entier de Gauss z est irréductible de type 1 ($z = 1 + i$), de type 2 ($z = x + iy$ où $x^2 + y^2$ est un nombre premier de $\mathbb{Z}$ congru à 1 modulo 4) ou de type 3 ($z \in \mathbb{Z}$ avec z congru à 3 modulo 4).

Démonstration

Soit $a + ib$ un entier de Gauss est irréductible

1. Supposons que $b = 0$ alors $a + ib = p \in \mathbb{Z}$
   Si p est irréductible, alors p est un nombre premier et p peut être :
   — congru à 3 modulo 4.
     Montrons qu'un nombre premier $p \in \mathbb{Z}$ congru à 3 modulo 4 est irréductible dans $\mathbb{Z}[i]$.
     Raisonnons par l'absurde et supposons que $p = (a_1 + ia_2)(b_1 + ib_2)$.
     En égalant le carré des modules on obtient :
     $p^2 = (a_1^2 + a_2^2)(b_1^2 + b_2^2)$ Comme p est premier cela entraine que :
     $a_1^2 + a_2^2 = p$ et que $(b_1^2 + b_2^2) = p$ puisque $(a_1 + ia_2)$ et $(b_1 + ib_2)$ ne sont pas inversibles, on sait que $a_1^2 + a_2^2 \neq 1$ et $(b_1^2 + b_2^2) = \neg 1$.
     On a donc obtenu une contradiction puisque la somme de 2 carrés n'est jamais congru à 3 modulo 4.
     Donc un nombre premier $p \in \mathbb{Z}$ congru à 3 modulo 4 est irréductible dans $\mathbb{Z}[i]$.

— congru à 2 modulo 4.
Montrons qu'un nombre premier $p \in \mathbb{Z}$ congru à 2 modulo 4 n'est pas irréductible dans $\mathbb{Z}[i]$.
Comme p est premier cela entraine que $p = 2$.
Puisque $p = 2 = (1 + i) * (1 - i)$, p n'est pas irréductible dans $\mathbb{Z}[i]$
— congru à 1 modulo 4.
Montrons qu'un nombre premier $p \in \mathbb{Z}$ congru à 1 modulo 4 n'est pas irréductible dans $\mathbb{Z}[i]$.
On sait que si p est un nombre premier congru à 1 modulo 4, il existe a et b dans $\mathbb{N}$ tels qie $p = a^2 + b^2$ donc $p = (a + ib) * (a - i * b)$ donc p n'est pas irréductible.

Donc si $p \in \mathbb{Z}$ est irréductible dans $\mathbb{Z}[i]$ c'est que p est premier et congru à 3 modulo 4.

2. Soit un nombre $a + ib$ irréductible dans $\mathbb{Z}[i]$ avec $b \neq 0$ ou $a \neq 0$. On en déduit que $g = gcd(a, b) = 1$ car sinon $a + i * b$ ne serait pas irréductible car on aurait $(a + ib) = g * (a/g + ib/g)$. Montrons tout d'abord que $p = a^2 + b^2$ est un nombre premier de $\mathbb{Z}$.
On a $p = a^2 + b^2 = (a + ib)(a - ib)$ avec $a + ib$ irréductible dans $\mathbb{Z}[i]$ donc $(a + ib)$ divise p_0 un diviseur premier de p dans $\mathbb{Z}$. Il existe donc il existe c et d dand $\mathbb{N}$ tels que :
$p_0 = (a + ib) * (c + id)$ avec $c^2 + d^2 \neq 1$.
En égalant le carré des modules on obtient :
$p_0^2 = (a^2 + b^2)(c^2 + d^2)$ où p_0 est un nombre premier de $\mathbb{Z}$ comme $c^2 + d^2 \neq 1$ et $(a^2 + b^2) \neq 1$ (puisque $b \neq 0$ ou $a \neq 0$), on a $p_0 = a^2 + b^2 = c^2 + d^2$.
Donc si $a + ib$ est irréductible dans $\mathbb{Z}[i]$ avec $b \neq 0$ ou $a \neq 0$ alors $(a^2 + b^2)$ est premier.
Donc $a^2 + b^2$ est congru à 1 ou 2 modulo 4.
Si $a^2 + b^2$ est congru à 2 modulo 4 c'est que $a^2 + b^2 = 2$ (car $(a^2 + b^2)$ est premier) et alors $a = 1$ et $b = 1$ sinon $a^2 + b^2$ est congru à 1 modulo 4.

Exercice
Écrire un programme `estirreductible`, qui prend en entrée un entier de Gauss z et retourne son type (1, 2 ou 3) si z est irréductble, et `0` sinon.

```
estirreductible(z):={
local p,m,n;
m:=abs(re(z));
n:=abs(im(z));
//if(type(m)!=2 or type(n)!=2){return faux; }:
if (z*conj(z)==1){return 0};
if (m==0) {m:=n;n:=0;}
if (n==0) {if (m>2 and irem(m-3,4)==0){return 3; }};
if (m==1 and n==1) {return 1; };
p:=m^2+n^2;
if (est_premier(p) and irem(p-1,4)==0){return 2; } else {return 0;}
}
:;
```

Exercice
Écrire un programme qui décompose un entier de Gauss en un produit de facteurs irréductbles.

```
decompose(z):={
local p,L1,L2,L3,s2,s0,s1,s3,k,c,lc,fc,j,z1,d;
L1:=NULL;
L3:=NULL;
//if (im(z)==0) {p:=abs(z);} else {p:=z*conj(z);}
p:=z*conj(z);
L2:=ifactors(p);
s2:=size(L2);
c:=L2[0];
s0:=0;
if (c==2) {L1:=L1,1+i,,L2[1];s0:=2;z:=z/(1+i);}
for (k:=s0;k<s2;k:=k+2) {
c:=L2[k];d:=L2[k+1];
if (irem(c-1,4)==0) {
   lc:=pa2b2(c);
   fc:=lc[1]+i*lc[0];
   j:=0;z1:=z;
   while(irem(z1,fc)==0)
      {j:=j+1;
        z1:=z1/fc;
       }
      if (j!=0 and j!=d) {
      L1:=L1,fc,j,conj(fc),d-j;
      z:=z/fc^j;z:=z/conj(fc)^(d-j);}
else {
     if (j==0) {L1:=L1,conj(fc),d;z:=z/conj(fc)^(d-j)}
else {L1:=L1,fc,j;z:=z/fc^j;}}
}
if (irem(c-3,4)==0) {L3:=L3,c,d/2;z:=z/c^(d/2);}
}
if (z==1){
 return [L1,L3];}
 else
{return [z,1,L1,L3];}
}
:;
```

Ou bien, on utilise la commande `ifactors` ou la commande `ifactor`.

7.3 Les nombres décomposables

Définition
On dit que $n \in \mathbb{Z} - \{-1, 0, 1\}$ est **décomposable**, si pour tout diviseur premier p de $1 + n^2$ il existe $d \in \mathbb{N}$ vérifiant $d < abs(n)$ et tel que $1 + d^2$ soit un multiple de

p.
Par exemple
8 est **décomposable** car $1+8^2=65=5*13$ et on a 5 divise $1+2^2=5$ ($d=2$) et 13 divise $1+5^2=26$ ($d=5$).
17 est **décomposable** car $1+17^2=290=2*5*29$ et on a 2 divise $1+1^2=2$ ($d=1$) et 5 divise $1+2^2=5$ ($d=2$) et 29 divise $1+12^2=145=29*5$ ($d=12$).
6 est **non décomposable** car $1+6^2=37=1*37$ et on a 1 divise $1+2^2=5$ ($d=2$) mais 37 ne divise pas $1+d^2$ quelque soit $0\leq d\leq 5$ (car 37> $(1+d^2)$ lorsque $0\leq d\leq 5$).
19 est **non décomposable** car $1+19^2=362=2*181$ et on a 2 divise $1+3^2=10$ ($d=2$) mais 181 ne divise pas $1+d^2$ quelque soit $0\leq d\leq 18$ (car 180 n'est pas un carré).
Exercice
Écrire une fonction `Xcas` `estdecomposable`, qui prend en entrée un entier n, et retourne le booléen `vrai` si n est d'ecomposable, `faux` sinon.
Et vérifier que les premiers entiers décomposables sont :
`3,7,8,12,13,17,18,21,23,27,30`

Les différentes solutions avec des algorithmes de plus en plus performents.

— **Algorithme 0**
L'**Algorithme 0** est un algorithme naif qui pour tout diviseur premier de $1+n^2$ cherche l'existence de d par balayage (la variable t sert de test).
Avec cet algorithme, si n est décomposable on trouve le plus petit entier d :
```
estdeconposable0(n):={
local p,Lf,k,p1,d,t,s;
n:=abs(n);
p:=1+n^2;
if (isprime(p)) {return faux};
Lf:=ifactors(p);
s:=size(Lf);
for (k:=0;k<s;k:=k+2){
p1:=Lf[k];
d:=1;
t:=0;
while(t!=1 and d<n) {
if (irem(1+d^2,p1)==0) {t:=1;afficher(d,p1);};
d:=d+1;
}
if (t==0){return faux;}
}
return vrai;
}:;
```
— **Algorithme 1**
L'**Algorithme 1** va faire appel aux nombres complexes.
$N=1+n^2$ est la norme au carrée de $z=1+i*n$ et soit p un diviseur premier de N.
Donc p divise $(1+in)(1-in)$.

— p n'est pas congru à 3 modulo 4.
En effet si p est congru à 3 modulo 4, p est irréductible dans $\mathbb{Z}[i]$ donc p divise $1 + in$ ou $1 - in$. Cela veut dire qu'il existe $x \in \mathbb{Z}$ et $y \in \mathbb{Z}$ tel que $p(x + iy) = 1 + in$ (ou $p(x + iy) = 1 - in$) ce qui est impossible car p diviserait 1 puisqu'en égalant les parties réelles on a $p * x = 1$
— p n'est pas congru à 0 modulo 4 car p est premier
— $p = 2$ ou p est congru à 1 modulo 4 En effet si p est congru à 2 modulo 4 c'est que $p = 2$ car p est premier

On sait que tout nombre premier p congru à 1 modulo 4 est la somme des carrés de 2 entiers donc il existe $a \in \mathbb{N}$ et $b \in N$:
$p = a^2 + b^2$.
On cherche les entiers d qui sont tels que :
$p = a^2 + b^2 = (a + i * b)(a - i * b)$ divise $1 + d^2 = (1 + i * d)(1 - i * d)$ donc
on cherche les entiers $d \in Z$ qui sont tels que :
$a + i * b$ divise $1 + i * d$ c'est à dire tel qu'il existe u et v vérifiant :
$(a + ib)(u + iv) = 1 + i * d$
donc $au - bv = 1$ et $av + bu = d$
Tous les d sont congrus modulo $p = a^2 + b^2$ (car si $au0 - bv0 = 1$ et $au - bv = 1$ alors $u = u0 + k * b$ et $v = v0 + k * a$)

```
estdecomposable1(n):={
local p,N,Lf,k,d,d1,a,b,L,a1,b1,s;
n:=abs(n);
N:=1+n^2;
if (isprime(N)) {return faux};
Lf:=ifactors(N);
s:=size(Lf);
for (k:=0;k<s;k:=k+2){
p:=Lf[k];
if (p!=2){
//on irem(p,4)==1 et p premier
a,b:=pa2b2(p);
if (a!=1 and b!=1) {
L:=iegcd(a,b);
a1:=L[0];
b1:=-L[1];
d:=irem(b*a1+a*b1,p);
d1:=irem(-b*a1-a*b1,p);
//afficher(d,d1,p);
};
if (d>=n et d1>=n) {return faux;}
}
}
return vrai;
}:;
```

— **Algorithme 2**

l'**Algorithme 2** simplifie l'algorithme1 puisque n et $-n$ se trouve être un d particuler, donc le plus petit d est égal à $n \bmod p$ ou à $-n \bmod p$.

```
estdecomposable2(n):={
local p,N,Lf,k,d,a,b,d1,s;
n:=abs(n);
N:=1+n^2;
if (isprime(N)) {return faux};
Lf:=ifactors(N);
s:=size(Lf);
for (k:=0;k<s;k:=k+2){
p:=Lf[k];
if (p!=2){
d:=irem(n,p);
d1:=irem(-n,p);
afficher(d,d1,p);}
if (d1>=n et d >=n){return faux;}
}
}
return vrai;
}:;
```

— **Algorithme**

L'**Algorithme** (le bon ! ! ! !) utilise le théorme suivant :

Théorème 1 :

Si p est premier et est congru à 1 modulo 4 alors il existe a et b tel que $p = a^2 + b^2$ et il existe $u + iv$ unique tel que :

$(a + ib)(u + iv) = 1 + im$ avec $m \in \mathbb{Z}$, $abs(m) < p/2$ et $u^2 + v^2 \leq 0.29p$

Cela entraine que si p est premier et congru à 1 modulo 4 alors il existe $m \in \mathbb{Z}$ tel que p divise $1 + m^2$ et m est unique dans $]-p/2; p/2[$.

Démonstration

Soit $p \in \mathbb{N}$ un nombre premier congru à 1 modulo 4 il existe a et b tel que $p = a^2 + b^2$.

On cherche $u, v, m \in \mathbb{Z}^3$ de valeur absolue la plus petite possible tels que :

$(a + ib)(u + iv) = 1 + im$

on a donc :

$au - bv = 1$ et $av + bu = m$. Puisque $p = a^2 + b^2$ est premier a et b sont premier entre eux donc il existe u et v (identité de Bézout) tels que :

$au - bv = 1$ et les solutions sont de la forme :

$u = u_0 + kbv = v_0 + ka$ $k \in \mathbb{Z}$ où u_0, v_0 est une solution particulière de $au - bv = 1$.

On a donc :

$(a + ib)(u0 + iv0 + k(b + ia)) = 1 + im$

$(a + ib)(u0 + iv0) + ik(a^2 + b^2) = 1 + im$

Donc : $m = av_0 + bu_0 + k(a^2 + b^2)$ $k \in \mathbb{Z}$ c'est à dire les m solutions sont ègaux modulo p.

Comme $a^2 + b^2 = p$ est impair on en déduit qu'il existe un m_0 et un seul tel que $|m_0| < p/2$ ($m \in \{(1 - p)/2, .. - 1, 0, 1...(p - 1)/2\}$).

Comme $(a+ib)(u+iv) = 1+im_0$, on a $(a^2+b^2)(u^2+v^2) = 1+m_0^2 = p(u^2+v^2)$ Puisque p est un nombre premier congru à 1 modulo 4, on a $p \geq 5$ donc $p^2 \geq 25$ ou encore $1/p \leq p/25$ donc :
$p(u^2+v^2) = 1+m_0^2 < 1+p^2/4$ donc
$(u^2+v^2) \leq 1/p + p/4 < p/25 + p/4 = 0.29p$
On tape :

```
estdecomposable(n):={
local p,N,Lf,k,d,d1,s,s1;
n:=abs(n);
N:=1+n^2;
if (isprime(N)) {return faux};
Lf:=ifactors(N);
s:=size(Lf);
p:=Lf[0];
if (p==2){s1:=2} else {s1:=0};
for (k:=s1;k<s;k:=k+2){
p:=Lf[k];
//on a irem(p,4)==1 et p premier
if (abs(n)<p/2) {return faux;}
}
return vrai;
}:;
```

On tape pour avoir la liste des nombres décomposables plus petit que n :

```
Lestdecomposable(n):={
local L,k;
L:=NULL;
for (k:=1;k<=n;k++){
if (estdecomposable(k)) {
L:=L,k;
}
}
return L;
}:;
```

```
Lestdecomposable(35)
```

On obtient :

```
3,7,8,13,17,18,21,30,31,32
```

7.4 Les nombres complétables

Definition :
On dit que le nombre $z = a+ib$ est complétable si il existe $u+iv \in \mathbb{Z}[i]$ et $m \in \mathbb{Z}$ tel que : $(a+ib)(u+iv) = 1+im$.
Théorème2 :
Si a et b sont premiers entre eux alors $z = a+ib$ est complétable. De plus, si $a \neq 1$ et $b \neq 1$, il existe $u, v, m \in \mathbb{Z}^3$ uniques tel que $(a+ib)(u+iv) = 1+im$ et $|m| < (a^2+b^2)/2$ et $u^2+v^2 < 0.29(a^2+b^2)$.

De plus u et v sont premiers entre eux et donc $u + iv$ et $u - iv$ sont complétables.
Démonstration
C'est le théorème1 sans l'hypothèse $p = a^2 + b^2$ premier.
Tout d'abord $1 + in$ et $n + i$ sont complétables car $(1 + in) * 1 = 1 + in$ et $(n + i) * (-i) = 1 - in$ et on a $n < (1 + n^2)/2$ si et seulement si $n \neq 1$.
Soient $a \neq 1$ et $b \neq 1$ premiers entre eux et $z = a + ib$.
On suppose donc que $a^2 + b^2 \geq 5$.
On a vu (cf th 1) que les m vérifiant $(a+ib)(u+iv) = 1+im$ sont congru modulo $a^2 + b^2$.
On va considérer 2 cas :

— $a^2 + b^2$ est impair i.e. $a^2 + b^2 = 2q + 1$
  il existe donc un m unique dans $\{-q, -q + 1... - 1, 0, 1, ..q\}$ donc
  $m \leq q < q + 1/2 = (a^2 + b^2)/2$
— $a^2 + b^2$ est pair i.e. $a^2 + b^2 = 2q$ et $q > 2$
  il existe donc un m unique dans $\{-q + 1... - 1, 0, 1, ..q\}$.
  Mais on ne peut pas avoir $m = q$ car alors on aurait :
  $(a^2 + b^2)(u^2 + v^2) = 2q(u^2 + v^2) = 1 + q^2$ donc q diviserait 1.
  Donc il existe un m unique dans $\{-q + 1... - 1, 0, 1, ..q - 1\}$ donc :
  $m \leq q - 1 < q = (a^2 + b^2)/2$.

Puisque $a^2 + b^2 \geq 5$ on a $(a^2 + b^2)^2 \geq 25$ ou encore :
$1/(a^2 + b^2) \leq (a^2 + b^2)/25$ donc :
$(a^2 + b^2)(u^2 + v^2) = 1 + m^2 < 1 + (a^2 + b^2)^2/4$ donc
$(u^2 + v^2) < 1/(a^2 + b^2) + (a^2 + b^2)/4 \leq (1/25 + 1/4)(a^2 + b^2) = 0.29(a^2 + b^2)$.

Le programme `complete` ci-dessous a comme argument $z = a+in$ et renvoie $u+iv, 1+im$ avec $(a+ib)(u+iv) = 1+im$ et $-(a^2+b^2)/2 < m < (a^2+b^2)/2$

```
complete(z):={
local p,m,a,b,u,v,L;
a:=re(z);
b:=im(z);
p:=a^2+b^2;
if (gcd(a,b)!=1){ return faux;}
if (a==1) {return 1,1+i*b};
if (b==1) {return -i,1-i*a};
L:=iegcd(a,b);
u:=L[0];
v:=-L[1];
m:=irem(b*u+a*v,p);
if (m>p/2){m:=m-p;}
return u+i*v,1+i*m;
}:;
```

Si on veut une liste des nombres qui complétent le nombre complétable $a + ib$, on tape :

```
Lcomplete(z):={
local L,z1,a1,b1,d1,k,b,a ;
L:=complete(z);
```

```
z1:=L[0];
a1:=re(z1);
b1:=im(z1);
a:=re(z);
b:=im(z);
d1:=L[1];
return ([a1+k*b+i*b1+i*k*a,d1+i*k*abs(z)^2])$(k=-5..5);
}:;
```

7.5 L'algorithme de Todd

7.5.1 Le Théorème3

Soit $n \geq 2$ un entier décomposable (i.e. pour tout diviseur premier p de $1+n^2$ et il existe d, $0 < d < n$, tel que p soit aussi un diviseur premier de $1+d^2$).
Il existe un entier $M > 1$ et des entiers w_j ($|w_j| < n$) tels que :
$M(1+in) = \epsilon(1+iw_1)^{a_1}(1+iw_2)^{a_2}...(1+iw_k)^{a_k}$ ($\epsilon \in \{1,-1,i,-i\}$).
Cette formule donne en égalant les arguments :
atan$(n) = arg(\epsilon) + \sum_{j=1}^{k} a_j$atan$(w_j) + m\pi$.
et puisque pour $x > 0$ on a :
atan(x) + atan$(1/x) = \pi/2$
on a aussi :
atan$(1/n) = -arg(\epsilon) + \sum_{j=1}^{k}$ atan$(1/w_j) + q * \pi$.

Soit $n \geq 2$ un entier décomposable. L'algorithme de Todd factorise $1+in$ et compléte par $u+iv$ chaque facteur irréductible $a+ib$ si $a \neq 1$ ou $b \neq 1$ pour avoir $(a+ib)(u+iv) = 1+im$
Par exemple si $(1+in) = (a+ib)*z1$, on obtient :
$(1+in)(u+iv)(u-iv) = (a+ib)(u+iv)(u-iv)z1$ donc
$(1+in)(u^2+v^2) = (1+im)(u-iv)z1$
Si $u == 1$ ou $v == 1$ on s'occupe d'un autre facteur irréductible, sinon on compléte $(u+iv)$....
Comme $(u^2+v^2) < 0.29(a^2+b^2)$ u^2+v^2 diminue à chaque complétion, donc l'algorithme s'arrête.

7.5.2 Un Exemple

On va montrer sur un exemple comment fonctionne l'algorithme de Todd.
Soit $n = 342$ n est décomposable.
On tape :
`ifactors(1+i*342)`
On obtient :
`[-1,1,2-i,1,10-7*i,1,11-6*i,1]`
On a : $-(2-i) = -2+i = i(1+2i)$
Donc $1+342i = i(1+2i)(10-7i)(11-6i)$
On tape :
`complete(10-7*i)`

On obtient :
`-2+3*i,1+44*i`
donc $(10-7i)*(-2+3i) = 1+44i$
On tape :
`complete(-2-3*i)`
On obtient :
`1+i,1-5*i`
donc
$(10-7i)(-2+3i)(-2-3i)(1+i)(1-i) = 26(10-7i) = (1+44i)(1-5i)(1-i)$
On tape :
`complete(11-6*i)`
On obtient :
`-1+2*i,1+28*i`
donc $(11-6*i)(-1+2i) = 1+28i$
donc $(11-6*i)(-1+2i)(-1-2i) = 5(11-6*i) = (1+28i)(-1-2i)$
Donc puisque $-(2-i) = i(1+2i)$ on a :
$5*26(1+342i) = i(1+2i)(1+44i)(1-5i)(1-i)(1+28i)(-1-2i)$
$130(1+342i) = -i(1-i)(1+2i)^2(1-5i)(1+28i)(1+44i)$

7.5.3 Exercice

1. Soit $f(x) = \text{atan}(x) + \text{atan}(\frac{1}{x})$.
   Déterminer le domaine de définition de f.
   Calculer $f'(x)$ puis montrer que :
   $$\text{atan}(x) + \text{atan}(\frac{1}{x}) = \frac{\pi}{2} \text{ si } x > 0$$
   $$\text{atan}(x) + \text{atan}(\frac{1}{x}) = -\frac{\pi}{2} \text{ si } x < 0$$
2. Soit $g(x) = \text{atan}(\frac{x+a}{1-ax}) - \text{atan}(x)$ pour $a \in \mathbb{R}$ Déterminer le domaine de définition de g.
   Calculer $g'(x)$ puis montrer que :
   si $ab < 1$ alors $\text{atan}(a) + \text{atan}(b) = \text{atan}((a+b)/(1-ab))$ et
   si $ab = 1$ et $a > 0$ alors $\text{atan}(a) + \text{atan}(b) = \pi/2$ et
   si $ab = 1$ et $a < 0$ alors $\text{atan}(a) + \text{atan}(b) = -\pi/2$.
   si $ab > 1$ et $a > 0$ alors $\text{atan}(a) + \text{atan}(b) = \pi + \text{atan}((a+b)/(1-ab))$ et
   si $ab > 1$ et $a < 0$ alors $\text{atan}(a) + \text{atan}(b) = -\pi + \text{atan}((a+b)/(1-ab))$.
3. En déduire que :
   $\text{atan}(28) + \text{atan}(44) = \pi + \text{atan}(-72/1231)$
   $2\text{atan}(2) + \text{atan}(-5) = \pi + \text{atan}(-4/3) + \text{atan}(-5) = \text{atan}(19/17)$
   $\pi + \text{atan}(-72/1231) + \text{atan}(19/17) = \pi + \text{atan}(341/343)$ donc
   $atan(-1) + 2\text{atan}(2) - \text{atan}(5) + \text{atan}(28) + \text{atan}(44) = \pi + atan(-1) + \text{atan}(341/343) =$
   $\pi - \text{atan}(1/342) = \pi/2 + \text{atan}(342)$
   Donc puisque $\text{atan}(-1) = -\pi/4$ et puisque pour $x > 0$ on a :
   $\text{atan}(x) + \text{atan}(1/x) = \pi/2$ on obtient :
   $\text{atan}(342) = -3\pi/4 + 2\text{atan}(2) - \text{atan}(5) + \text{atan}(28) + \text{atan}(44)$

Donc puisque pour $x > 0$ on a $\text{atan}(x) + \text{atan}(1/x) = \pi/2$
$\text{atan}(1/342) = -\pi/4 + 2\text{atan}(1/2) - \text{atan}(1/5) + \text{atan}(1/28) + \text{atan}(1/44)$

7.5.4 Un programme qui simplie les formules précédentes

On peut écrire un petit programme `ajuste` pour simplifier : $\sum_{j=1}^{k} a_j \text{atan}(w_j)$. `ajuste` a comme argument la liste formée par les arguments w_j des arcs tangentes en répétant a_j fois w_j et renvoie `C,a` tel que :
$\sum_{j=1}^{k} a_j \text{atan}(1/w_j) = C + \text{atan}(a)$.
Par exemple pour la formule :
$\text{atan}(-1) + 2\text{atan}(2) - \text{atan}(5) + \text{atan}(28) + \text{atan}(44)$. on met comme argument la liste `L:=[-1,2,2,-5,28,44]` et `ajuste(L)` renvoie `pi,-1/342`.
On définit f la fonction f par :
`f(a,b):=(a+b)/(1-a*b);`
Pour simplifier $\text{atan}(a) + \text{atan}(b)$, on écrit tout d'abord la fonction `simpl(a,b)` qui renvoie `C,c` tel que :
$\text{atan}(a) + \text{atan}(b) = C + \text{atan}(c)$
Puis, on écrit `ajuste(L)` qui renvoie `C,c` tel que :
$\sum_j \text{atan}(L[j]) = C + \text{atan}(c)$

```
f(a,b):=(a+b)/(1-a*b);
simpl(a,b):={
local s;
if (a==0) {return 0,b;}
if (b==0) {return 0,a;}
s:=sign(a);
if (a*b==1) {return s*pi/2,0;}
if  (a*b<1) {return 0,f(a,b);}
if  (a*b>1) {return s*pi,f(a,b);}
}:;
ajuste(L):={
local s,a,b,C,c,ajou,k;
ajou:=0;
s:=size(L);
a:=L[0];
k:=1;
while (k<s){
b:=L[k];
C,c:=simpl(a,b);
ajou:=ajou+C;
a:=c;
k:=k+1;
}
return normal(ajou),a;
}:;
```

On tape : `ajuste([-1,2,2,-5,28,44])`
On obtient : `pi,-1/342`
Donc :

$\text{atan}(342) + \pi/2 = \text{atan}(-1/342) + \pi = \text{atan}(-1) + 2 * \text{atan}(1/2) - \text{atan}(1/5) + \text{atan}(1/28) + \text{atan}(1/44)$
Donc :
$\text{atan}(342) = -\pi/2 + \text{atan}(-1) + 2 * \text{atan}(2) - \text{atan}(5) + \text{atan}(28) + \text{atan}(44)$
On tape : `ajuste([-1,1/2,1/2,-1/5,1/28,1/44])`
On obtient : `0,1/342`
Donc :
$\text{atan}(1/342) = \text{atan}(-1) + 2 * \text{atan}(1/2) - \text{atan}(1/5) + \text{atan}(1/28) + \text{atan}(1/44)$

Remarque
En utilisant `Lcomplete(11-6*i)` on voit que :
$(11 - 6i)(5 - 9i) = 1 - 129i$
On peut aussi utiliser cette complétion car $123 < 342/2$ et $5^2+9^2 = 106 < 342/3$.
On tape :
`complete(5+9i)`
On obtient :
`2+i,1+23*i`
Donc $(5 + 9i)(2 + i) = 1 + 23i$
Donc on a :
$(11-6i)(5-9i)(5+9i)(2+i)(2-i) = 5*106(11-6i) = (1-129i)(1+23i)(2-i)$
On obtient alors :
$1 + 342i = i(1 + 2i)(10 - 7i)(11 - 6i)$
$5*106*26(1+342i) = i(1+2i)(1+44i)(1-5i)(1-i)(1-129i)(1+23i)(2-i)$
$13780(1 + 342i) = (1 - i)(1 + 2i)^2(1 - 5i)(1 + 23i)(1 + 44i)(1 - 129i)$
Donc :
$\text{atan}(342) = -\pi/4 + 2\text{atan}(2) - \text{atan}(5) + \text{atan}(23) + \text{atan}(44) - \text{atan}(129)$
Donc puisque pour $x > 0$ on a $\text{atan}(x) + \text{atan}(1/x) = \pi/2$ on a :
$\text{atan}(1/342) = -\pi/4 + 2\text{atan}(1/2) - \text{atan}(1/5) + \text{atan}(1/23) + \text{atan}(1/44) - \text{atan}(1/129)$
On tape : `ajust([-1,2,2,-5,23,44,-129])`
On obtient : `0,1/342`
On tape : `ajust([-1,1/2,1/2,-1/5,1/23,1/44,-1/129])`
On obtient : `0,-342`
Traduction de l'algorithme de Todd en langage `Xcas`

```
todd(n):={
 local L,k,s,s1,z,R,Z,Zc,zc,p;
 //if (estcompletable(n)==faux) {return faux;};
 L:=ifactors(1+i*n);
 s:=size(L);
 R:=NULL;
 p:=1;
 for (k:=2;k<s;k:=k+2){
  z:=L[k];
  if (im(z)^2==1 or re(z)^2==1) {
    R:=R,z,L[k+1];}
  else {
    Z:=complete(z);
```

```
    R:=R,Z[1],L[k+1];
    zc:=conj(Z[0]);
    p:=p*abs(zc)^(2*L[k+1]);
    while (re(zc)^2!=1 and im(zc)^2!=1){
      Zc:=complete(zc);
      R:=R,Zc[1],L[k+1];
      zc:=conj(Zc[0]);
      p:=p*abs(zc)^(2*L[k+1]);
    }
    R:=R,zc,L[k+1];
  }
 }
 return normal(p),[L[0],1,R];
}:;
```

On tape :
`todd(342)`
On obtient :
`130,[-1,12-i,1,1+44*i,1,1-5*i,1,1-i,1,1+28*i,1,-1-2*i,1]`
On tape :
`todd(266)`
On obtient :
`1850,[i,1,1-80*i,1,1+6*i,1,1-143*i,1,-1+7*i,1]`

7.6 Pour obtenir une formule de type Machin

Une formule de type Machin est une formule qui ressemble à :

$$\text{atan}(1/n) = k * \pi + \sum_{j=0}^{k} a_j \text{atan}(1/wj)$$

avec $0 < wj < n$ Si le but est d'obtenir une formule avec des arcs tangentes, on modifie l'algorithme précédent pour avoir seulement comme résultat la liste des w_j compté avec leur multiplicité.
On tape :

```
todd2(n):={
local L,k,s,s1,z,R,Z,Zc,zc,p,RR;
//if (estcompletable(n)==faux) {return faux;};
L:=ifactors(1+i*n);
s:=size(L);
R:=NULL;
p:=1;
for (k:=2;k<s;k:=k+2){
z:=L[k];
if (im(z)^2==1 or re(z)^2==1) {R:=R,z,L[k+1];}
else {Z:=complete(z);
R:=R,Z[1],L[k+1];
```

```
zc:=conj(Z[0]);
p:=p*abs(zc)^(2*L[k+1]);
while (re(zc)^2!=1 and im(zc)^2!=1){
Zc:=complete(zc);
R:=R,Zc[1],L[k+1];
zc:=conj(Zc[0]);
p:=p*abs(zc)^(2*L[k+1]);
}
R:=R,zc,L[k+1];
}
}
RR:=NULL;
s:=size(R);
for (k:=0;k<s;k:=k+2){
if (re(R[k])^2!=1){R[k]:=i*R[k];}
RR:=RR,(re(R[k])*im(R[k]))$(R[k+1]);
}
return sort(RR);
}
:;
```

On tape :
`todd2(342)`
On obtient :
`-5,-1,2,2,28,44`
On tape :
`ajust(todd2(342))`
On obtient :
`0,1/266`
Donc :
$\mathrm{atan}(1/342) = \mathrm{atan}(-1/5) + \mathrm{atan}(-1) + 2\mathrm{atan}(1/28) + \mathrm{atan}(1/44)$
On tape :
`todd2(266)`
On obtient :
`-143,-80,-7,6`
On tape :
`ajust(todd2(266))`
On obtient :
`0,1/266`
Donc :
$\mathrm{atan}(1/266) = \mathrm{atan}(-1/143) + \mathrm{atan}(-1/80) + \mathrm{atan}(-1/7) + \mathrm{atan}(1/6)$

Chapitre 8

Pour s'amuser avec le tableur de `Xcas`

8.1 Quiz

Vous venez d'être embauché à la Banque de France pour 3 ans.
On vous demande de choisir entre :
1/ Avoir une augmentation de 100 euros par mois à chaque début de semestre,
2/ Avoir une augmentation de 180 euros par mois à chaque début d'année.
Que préférez-vous ?
Pourquoi ? Justifier.
ou bien
On vous demande de choisir entre :
1/ Avoir une augmentation de 100 euros par mois à chaque début de semestre,
2bis/ Avoir une augmentation de 220 euros par mois à chaque début d'année.
Que préférez-vous ?
Pourquoi ? Justifier.
la solution sans le tableur Soient :
`G1(n)` le gain obtenu en euros à la fin de l'année `n` si on opte pour la solution 1 et
`G2(n)` le gain obtenu en euros à la fin de l'année `n` si on opte pour la solution 2.
On a `G1(1)=6*100=600` **la solution avec le tableur**

8.2 Nombre de carrés dans un échiquier

On cherche le nombre de carrés $NC(n)$ que l'on peut former sur un échiquier de dimension $n * n$.
On veut faire une fiche de travail pour que les élèves devine et démontre que :

$$NC(n) = 1^2 + 2^2 + 3^2 + ... + n^2 = \frac{n(n+1)(2n+1)}{6}$$

8.2.1 L'énoncé

On rappelle que $1 + 2 + ...n = \frac{n(n+1)}{2}$.

1. Combien peut-on former de carrés de côtés égal à 1 sur un échiquier de dimension $p * p$?

2. Combien peut-on former de carrés de côtés égal à n sur un échiquier de dimension $n * n$?
3. Expliquer pourquoi le nombre de carrés de côtés égal à $n-1$ sur un échiquier de dimension $n * n$ est égal au nombre de carrés de côtés égal à 1 sur un échiquier de dimension $2 * 2$.
4. Expliquer pourquoi le nombre de carrés de côtés égal à $n-2$ sur un échiquier de dimension $n * n$ est égal au nombre de carrés de côtés égal à 1 sur un échiquier de dimension $3 * 3$?
5. Sur le tableur de `Xcas` remplir :
   — la première colonne avec les nombres de 0 à n (par exemple $n = 39$).
   — la deuxième colonne avec le nombre de carrés de côtés égal à 1, sur un échiquier de dimension $p * p$ ($p = 1..n$) qui est aussi le nombre de carrés de dimension $n - p + 1$ sur un échiquier de dimension $n * n$.
   — la troisième colonne avec avec les sommes partielles de la deuxième colonne. Dire pourquoi cette colonne donne $NC(k)$ lorsque $A(k) = k$.
   — la quatrième colonne avec les sommes partielles de la première colonne.
   — la cinquième colonne avec le quotient de la la troisième colonne par la quatrième colonne.
   — Deviner l'expression de la cinquième colonne et en déduire la valeur de l'expression de $NC(n)$.

8.2.2 La correction avec `Xcas`

1. On a $0, 1, 2, 3..n$ sur la colonne A
2. On a $0, 1, 4, 9...n^2$ sur la colonne B
3. On a $0, 1, 5, 14, 30....$ sur la colonne C
4. On a $0, 1, 3, 6, 10, 15....$ sur la colonne D c'est à dire $k(k + 1)/2$ pour $k = 1..n$. Donc $D(k) = 1^2 + 2^2 + ...k^2 = NC(k)$.
5. On a $0, 1, 5/3, 7/3, 3, 11/3, 13/2, 5....$ sur la colonne E.
   On devine que la colonne E vaut $(2k + 1)/3$ pour $k = 1..n$.
   Puisque la colonne D vaut $k(k + 1)/2$ pour $k = 1..n$, on en déduit que $NC(k) = k(k + 1)/2 * (2k + 1)/3 = k(k + 1)(2k + 1)/6$.
   On peut maintenant montrer par récurrence que :
   $$1^2 + 2^2 + ...k^2 = \frac{k(k + 1)(2k + 1)}{6}$$
   car
   $1^2 = (1 * 2 * 3)/6$ et
   $$1^2 + ...k^2 + (k+1)^2 = \frac{(k + 1)(2k^2 + k + 6k + 6)}{6} = \frac{(k + 1)(k + 2)(2k + 3)}{6}$$

8.3 Étude d'une suite

8.3.1 L'énoncé

On considère la suite récurrente :
$u_0 = a$

$u_1 = b$
pour $n > 1$ $u_n = |u_{n-1}| - u_{n-2}$
Étudiez cette suite.
Faites des essais avec le tableur prendre par exemple :
$a = 3$ avec $b = 4, b = 5, b = 7$ etc...
$a = -3$ avec $b = 4, b = 5, b = 7$ etc...
Que remarquez-vous ?

8.3.2 La démarche mathématique

On va montrer que cette suite est périodique ($u_9 = u_0$ et $u_{10} = u_1$).
Pour cela on considère 9 cas :
6 cas lorsque $b >= 0$ et 3 cas lorsque $b < 0$:
cas 1 : $b \geq 2a \geq 0$
On obtient :
`a,b,b-a,-a,-b+2a,b-a,2b-3a,-2a+b,a-b,a,b....`
cas 2 : $2a > b \geq a$
On obtient :
`a,b,b-a,-a,-b+2a,3a-b,a,2a+b,a-b,a,b.....`
cas 3 : $a/2 > b \geq 0$
On obtient :
`a,b,b-a,a-2b,-3b+2a,a-b,2b-a,-b,-b+a,a,b....`
cas 4 : $a > b \geq a/2$
On obtient :
`a,b,b-a,a-2b,b,3b-a,2b-a,-b,a-b,a,b...`
cas 5 : $b \geq -a \geq 0$
On obtient :
`a,b,b-a,-a,-b,b+a,2b+a,b,b-a,a,b....`
cas 6 : $-a \geq b \geq 0$
On obtient :
`a,b,b-a,-a,-b,b+a,-a,-2a-b,-a-b,a,b....`
cas 7 : $-b \geq a > 0$
On obtient :
`a,b,-b-a,-2b-a,-b,b+a,-a,-b,-b+a,a,b....`
cas 8 : $a > -b > 0$
On obtient :
`a,b,-b-a,a,2a+b,a+b,-a,-b,-b+a,a,b....`
cas 9 : $-b > 0 \geq a$
On obtient :
`a,b,-b-a,-2b-a,-b,a+b,-a,-2a-b,-b-a,a,b....`

8.3.3 La correction avec `Xcas`

Avec `Xcas` on tape (cf **cas9**) :
`assume(a<0)` et la réponse est `a`
`assume(b<0)` et la réponse est `b`

`normal(abs(ans())-ans(-2))` et la réponse est `-b-a`
puis `enter`, `enter`... :
la réponse est `-2*b-a`
la réponse est `-b`
la réponse est `b+a`
la réponse est `-a`
la réponse est `-b-2*a`
la réponse est `-b-a`
la réponse est `a`
la réponse est `b`
Attention La commande `assume` ne doit utiliser que des noms de variables sans opérateur. Pour faire les différents cas avec `Xcas` il faut donc faire des changements de variables, par exemple pour traiter le **cas1** ($b \geq 2a \geq 0$), on pose $c = b - 2 * a$ donc $b = c + 2 * a$.
Avec `Xcas` on tape (cf **cas9**) :
`assume(a>0)` et la réponse est `a`
`assume(c>0)+2*a` et la réponse est `c+2*a`
`normal(abs(ans())-ans(-2))` et la réponse est `c+a`
puis `enter`, `enter`... :
la réponse est `-a`
la réponse est `-c`
la réponse est `c+a`
la réponse est `2*c+a`
la réponse est `c`
la réponse est `-c-a`
la réponse est `a`
la réponse est `c+2*a`
On peut aussi utiliser le tableur :
on tape (cf **cas9**) :
`assume(a>0);assume(c>0)`
Puis on ouvre le tableur, et on remplit :
`A0` avec `a`, `A1` avec `c+2*a`, `A2` avec `=normal(abs(A1)-A0)`
puis on appuie sur le bouton `remplir` (ou `fill`) et on peut voir les différentes valeurs de la suite.
On peut aussi avec le tableur mettre les 9 cas dans les colonnes `A`...`I`.

Il faut taper `purge(a)` si on a fait aupparavant `assume(a>0)`
Puis on tape : `assume(c>0);assume(d>0)`
et on pose :
cas1 ($b \geq 2a \geq 0$)
$\mathtt{A0} = \mathtt{d}, \mathtt{A1} = \mathtt{c} + 2 * \mathtt{d}$ (ie $a = d, c = b - 2a$)
cas2 ($2a \geq b \geq a$)
$\mathtt{B0} = \mathtt{d} + \mathtt{c}, \mathtt{B1} = \mathtt{c} + 2 * \mathtt{d}$ (ie $d = b - a, c = -b + 2a$)
cas3 ($a/2 > b \geq 0$)
$\mathtt{C0} = \mathtt{d} + 2\mathtt{c}, \mathtt{C1} = \mathtt{c}$ (ie $d = a - 2b, c = b$)
cas4 ($a > b \geq a/2$)
$\mathtt{D0} = 2 * \mathtt{d} + 2 * \mathtt{c}, \mathtt{D1} = 2 * \mathtt{c} + \mathtt{d}$ (ie $d = a - b, 2 * c = 2 * b - a$)
cas5 ($b \geq -a \geq 0$)
$\mathtt{E0} = -\mathtt{d}, \mathtt{E1} = \mathtt{c} + \mathtt{d}$ (ie $d = -a, c = b + a$)

cas6 $(-a \geq b \geq 0)$
$\texttt{F0} = -\texttt{c} - \texttt{d}, \texttt{F1} = \texttt{d}$ (ie $c = -a - b, b = d$)
cas7 $(-b \geq a > 0)$
$\texttt{G0} = \texttt{d}, \texttt{G1} = -\texttt{c} - \texttt{d}$ (ie $c = -a - b, d = a$)
cas8 $(a > -b > 0\,)$
$\texttt{H0} = \texttt{c} + \texttt{d}, \texttt{H1} = -\texttt{d}$ (ie $d = -b, c = a + b$)
cas9 $(-b > 0 \geq a)$
$\texttt{I0} = -\texttt{d}, \texttt{I1} = -\texttt{c}$ (ie $d = -a, b = -c$)
On met dans `A2` la formule `=normal(abs(A1)-A0)`
Puis on recopie cette formule vers le bas (bouton `remplir`) et sur le coté (menu `Edit` sous-menu `mtrw(Editeur de matrices)` et `copier->`).
On obtient :

	A	B	C	D	E	F	G	H	I
0	d	d+c	d+2*c	2*d+2*c	-d	-c-d	d	c+d	-d
1	c+2*d	c+2*d	c	2*c+d	c+d	d	-c-d	-d	-c
2	c+d	d	-d-c	-d	c+2*d	c+2*d	c	-c	c+d
3	-d	-c-d	d	-(2*c)	d	c+d	2*c+d	c+d	2*c+d
4	-c	c	2*d+c	2*c+d	-c-d	-d	c+d	2*c+d	c
5	c+d	2*c+d	d+c	4*c+d	c	-c	-c	c	-c-d
6	2*c+d	c+d	-d	2*c	2*c+d	c+d	-d	-c-d	d
7	c	-c	-c	-2*c-d	c+d	2*c+d	d+c	d	c+2*d
8	-c-d	-d	c+d	d	-c	c	2*d+c	c+2*d	c+d
9	d	d+c	2*c+d	2*c+2*d	-d	-c-d	d	c+d	-d
10	c+2*d	2*d+c	c	2*c+d	d+c	d	-d-c	-d	-c
11	c+d	d	-c-d	-d	2*d+c	c+2*d	c	-c	c+d

On remarque que les colonnes se déduisent les unes des autres :
si on considère la colonne `I` on a
`I1, I2` est égal à `F0, F1` (puisque `c` er `d` jouent le même rôle) de même,
`I2, I3` est égal à `B0, B1`,
`I3, I4` est égal à `C0, C1`,
`I4, I5` est égal à `G0, G1`,
`I5, I6` est égal à `F0, F1`,
`I6, I7` est égal à `A0, A1`,
`I7, I8` est égal à `D0, D1`,
`I8, I9` est égal à `H0, H1`.
On peut donc considérer la fonction f de $\mathbb{R}^2$ dans $\mathbb{R}^2$ définie par :
$f(x, y) = (y, |y| - x)$
On peut montrer que $f^9 = id$ en effet :
si $A0 = \{(x, y) \in \mathbb{R}^2, x < 0, y \leq 0\}$, on pose :
$A1 = f(A0)$, $A2 = f(A1) = f^2(A0)$... $A8 = f^8(A0)$
alors $A0, A1, A2.., A8$ forment une partition de $\mathbb{R}^2$,
de plus si $(a, b) \in A0$ alors $f^9(a, b) = (a, b)$.
Avec `Xcas` on tape :

```
f(L):={ return([L[1],normal(abs(L[1])-L[0])]);
};
assume(a<0);assume(b<=0)
f(f(f(f(f(f(f(f(f(([a,b])))))))))
```

On obtient : `[a,b]`
Cela prouve que la suite récurrente U définie par :
$(U_0, U_1) = (x, y)$
$(U_n, U_{n+1}) = f(U_{n-1}, U_n) = (U_n, |U_n| - U_{n-1})$ pour $n > 0$
est périodique de période 9 ($U_n = U_{n+9}$ pour tout $n \geq 0$)).
On peut visualiser en trois temps que `A0,A1...A8` forment une partition.
On tape :

```
for (j:=0;j<300;j:=j+1) {
  c:=rand(0..2);
  d:=rand(0..3);
  point(-c+i*(c+d));
  point(c+d+i*(2*c+d));
  point(2*c+d+i*c);
  point(c-i*(c+d));
}
```

et on voit les régions `A1,A2,A3,A4` sur l'écran de géométrie.
On tape :

```
for (j:=0;j<300;j:=j+1) {
  c:=rand(0..2);
  d:=rand(0..3);
  point(-c-d+i*d);
  point(d+i*(2*d+c));
  point(2*d+c+i*(c+d));
  point(c+d-i*d);
}
```

et on voit les régions `A5,A6,A7,A8` sur l'écran de géométrie.

Chapitre 9

Pour s'amuser avec le graphique

9.1 Les polygones et les milieux de leurs côtés

9.1.1 Le triangle et le quadrilatère

Le triangle

Étant donné 3 points `A,B,C`, construire un triangle `E,F,G` tel que `A` soit le milieu de `EF`, `B` soit le milieu de `FG` et `C` soit le milieu de `GE`.
Avec `Xcas`, faisons des essais : On clique sur 4 points `A,B,C,E` puis on tape :

```
F:=symetrie(A,E);
G:=symetrie(B,F);
H:=symetrie(C,G);
polygone(A,B,C);
polygone_ouvert(E,F,G,H);
```

On fait bouger ensuite le point `E` pour que `E` et `H` coincident. On analyse alors la figure : Lorsque `E` et `H` coincident `EG` est parallèle à `AB` et le vecteur `CE` est égal au vecteur `BA` (propriété des milieux d'un triangle).
On en déduit la construction avec `Xcas` : On clique sur 3 points `A,B,C` puis on tape :

```
E:=translation(A-B,C);
F:=symetrie(A,E);
G:=symetrie(B,F);
```

Le quadrilatère

Étant donné 4 points `A,B,C,D`, construire un quadrilatère `E,F,G,H` tel que `A` soit le milieu de `EF`, `B` soit le milieu de `FG`, `C` soit le milieu de `GH`, `D` soit le milieu de `HE`.
Avec `Xcas`, faisons des essais :
On clique sur 5 points `A,B,C,D,E` (il faut renommer les points car `D` n'est pas attribué automatiquement car en Maple `D` désigne la dérivation).

```
F:=symetrie(A,E);
G:=symetrie(B,F);
```

```
H:=symetrie(C,G);
I:=symetrie(D,H);
polygone(A,B,C,D);
polygone_ouvert(E,F,G,H,I);
```

On fait bouger ensuite le point `E` pour que `E` et `I` coincident. Mais, cette fois on n'y arrive pas ....On modifie le point `A` pour que `E` et `I` coincident. On analyse alors la figure : Lorsque `E` et `I` coincident `ABCD` est un parallèlogramme (on a $2\overrightarrow{AB} = \overrightarrow{EG}$ et $2\overrightarrow{DC} = \overrightarrow{IG}$ donc si `E` et `I` coincident on a $\overrightarrow{AB} = \overrightarrow{DC}$).
Lorsque `ABCD` est un parallèlogramme, on remarque alors que si on fait bouger le point `E`, on a toujours `E` et `I` en coincidence. En effet on a :
$2\overrightarrow{AB} = \overrightarrow{EG}$ et $2\overrightarrow{DC} = \overrightarrow{IG}$ donc si $\overrightarrow{AB} = \overrightarrow{DC}$, on a $\overrightarrow{EG} = \overrightarrow{IG}$ donc `E` et `I` coincident.

Analyse

Dans le cas du triangle, lorsqu'on fait subir au point `E`, 3 symétries centrales successives S_1, S_2, S_3 on veut retrouver `E` : cela signifie que `E` doit être un point fixe de $S_3 \circ S_2 \circ S_1$.

Dans le cas du quadrilatère, lorsqu'on fait subir au point `E`, 4 symétries centrales successives S_1, S_2, S_3, S_4 on veut retrouver `E` : cela signifie que `E` doit être un point fixe de $S_4 \circ S_3 \circ S_2 \circ S_1$. On est donc amené à comprendre comment on compse des symétries centrales.

9.1.2 Translation et composition de symétries centrales

On désigne par $\mathcal{S}_O$ la symétrie de centre O et par $\mathcal{T}_{AB}$ la translation de vecteur $\overrightarrow{AB}$.
Théorème Soient deux points O_1 et O_2 et un vecteur V, on a :

1. $\mathcal{S}_{O_2} \circ \mathcal{S}_{O_1}$=$\mathcal{T}_{2O_1O_2}$,
2. $\mathcal{T}_V \circ \mathcal{S}_{O_1}$=$\mathcal{S}_K$ avec $\overrightarrow{O_1K} = V/2$,
3. $\mathcal{S}_{O_1} \circ \mathcal{T}_V$=$\mathcal{S}_H$ avec $\overrightarrow{O_1H} = -V/2$.

En effet,

1. Soit A un point. Posons $B = \mathcal{S}_{O_1}(A)$ et $C = \mathcal{S}_{O_2}(B)$.
   On a : $\overrightarrow{AO_1} = \overrightarrow{O_1B}$ et, $\overrightarrow{O_2C} = \overrightarrow{BO_2}$ donc, $\overrightarrow{AC} = \overrightarrow{AO_1} + \overrightarrow{O_1O_2} + \overrightarrow{O_2C} = \overrightarrow{O_1B} + \overrightarrow{O_1O_2} + \overrightarrow{BO_2} =$
   $2\overrightarrow{O_1O_2}$
   Comme A est quelconque on en déduit que :
   $\mathcal{S}_{O_2} \circ \mathcal{S}_{O_1}$=$\mathcal{T}_{2O_1O_2}$
2. Soit A un point, V un vecteur et K tel que $\overrightarrow{O_1K} = V/2$.
   Posons $B = \mathcal{S}_{O_1}(A)$ et $C = \mathcal{T}_V(B)$.
   On a : $\overrightarrow{AO_1} = \overrightarrow{O_1B}$,
   $\overrightarrow{BC} = V$ et,
   $\overrightarrow{O_1K} = V/2 = \overrightarrow{BC}/2$.
   Donc
   $\overrightarrow{AK} = \overrightarrow{AO_1} + \overrightarrow{O_1K} =$

$\overrightarrow{AO_1} + \overrightarrow{BC}/2$ et $\overrightarrow{AC} = \overrightarrow{AB} + \overrightarrow{BC} =$
$2\overrightarrow{AO_1} + \overrightarrow{BC} = 2\overrightarrow{AK}$ soit
$\overrightarrow{AC} = \overrightarrow{AK} + \overrightarrow{KC} = 2\overrightarrow{AK}$ c'est à dire
$\overrightarrow{KC} = \overrightarrow{AK}$
donc $C = \mathcal{S}_K(A)$ Comme A est quelconque on en déduit que :
$\mathcal{T}_V \circ \mathcal{S}_{O_1}$=$\mathcal{S}_K$ avec $\overrightarrow{O_1K} = V/2$

3. Soit A un point, V un vecteur et H tel que $\overrightarrow{O_1H} = -V/2$.
   Posons $B = \mathcal{T}_V(A)$ et $C = \mathcal{S}_{O_1}(B)$.
   On a : $\overrightarrow{O_1H} = -V/2$, $\overrightarrow{AB} = V$ et $\overrightarrow{BO_1} = \overrightarrow{O_1C}$
   Donc
   $\overrightarrow{AH} = \overrightarrow{AB} + \overrightarrow{BO_1} + \overrightarrow{O_1H} =$
   $V + \overrightarrow{O_1C} - V/2 = \overrightarrow{O_1C} + V/2$ et
   $\overrightarrow{AC} = \overrightarrow{AB} + \overrightarrow{BC} =$
   $V + \overrightarrow{BC} = V + 2\overrightarrow{O_1C}$ soit
   $\overrightarrow{AC} = \overrightarrow{AH} + \overrightarrow{HC} = 2\overrightarrow{AH}$ c'est à dire
   $\overrightarrow{HC} = \overrightarrow{AH}$
   donc $C = \mathcal{S}_H(A)$ Comme A est quelconque on en déduit que :
   $\mathcal{S}_{O_1} \circ \mathcal{T}_V$=$\mathcal{S}_H$ avec $\overrightarrow{O_1H} = -V/2$

9.1.3 Le pentagone

Étant donné 5 points `A,B,C,D,E`, construire un pentagone `A1,A2,A3,A4,A5` tel que `A` soit le milieu de `A1A2`, `B` soit le milieu de `A2A3`,....,`E` soit le milieu de `A5A1`.
La construction du pentagone revient à déterminer $A1$ tel que :
$\mathcal{S}_{A_1} = \mathcal{S}_E \circ \mathcal{S}_D \circ \mathcal{S}_C \circ \mathcal{S}_B \circ \mathcal{S}_A$, puis à construire les points
$A_2 = \mathcal{S}_A(A_1)$, $A_3 = \mathcal{S}_B(A_2)$....
On a d'après le théorème précédent :
$\mathcal{S}_B \circ \mathcal{S}_A = \mathcal{T}_{2AB}$ $\mathcal{S}_D \circ \mathcal{S}_C = \mathcal{T}_{2CD}$ donc
$\mathcal{S}_D \circ \mathcal{S}_C \circ \mathcal{S}_B \circ \mathcal{S}_A = \mathcal{T}_{2(AB+CD)}$ et
$\mathcal{S}_E \circ \mathcal{S}_D \circ \mathcal{S}_C \circ \mathcal{S}_B \circ \mathcal{S}_A = \mathcal{S}_E \circ \mathcal{T}_{2(AB+CD)} = \mathcal{S}_{A_1}$ avec $\overrightarrow{EA_1} = \overrightarrow{BA} + \overrightarrow{DC}$
La construction avec `Xcas` .
On clique sur 5 points `A,B,C,D,E` (il faut renommer les points car `D` n'est pas attribué automatiquement car en Maple `D` désigne la dérivation).

```
polygone(A,B,C,D,E);
A1:=translation(A-B+C-D,E);
A2:=symetrie(A,A1);
A3:=symetrie(B,A2);
A4:=symetrie(C,A3);
A5:=symetrie(D,A4);
F:=symetrie(E,A5):;
polygone_ouvert(A1,A2,A3,A4,A5,F);
A1==F;
```

La réponse de `A1==F` est `1` ce qui signifie que la construction est correcte.

9.1.4 Le polygône ayant un nombre impair de côtés

La construction d'un polygône ayant un nombre impair de côtés à partir des milieux de ses côtés est possible puisque le produit d'un nombre impair de symétries centrales est une symétrie centrale.

9.2 Le théorème de Johnson

Soient 3 cercles C_A, C_B, C_C de centre A, B, C de même rayon R et ayant un point commun O (on a donc les trois centres A, B, C sont sur un cercle de centre O et de rayon R).
Soient :
— M l'intersection autre que O de C_A et C_B,
— N l'intersection autre que O de C_A et C_C,
— P l'intersection autre que O de C_C et C_B
alors, les trois points M, N, P sont sur un cercle de rayon R.

Lemme
Si deux cercles de même rayon R et de centre O_1 et O_2 se coupent en A e B alors le quadrilatère O_1, A, O_2, B est un losange.



En effet on a $O_1A = O_1B = O_2A = O_2B = R$.

Démonstration
D'après le lemme les quadrilatères :
O, B, M, A,
O, A, N, C, et
O, C, P, B
sont des losanges.
La translation de vecteur $\overrightarrow{OA}$ transforme $\overrightarrow{BC}$ en $\overrightarrow{MN}$.
La translation de vecteur $\overrightarrow{OB}$ transforme $\overrightarrow{AC}$ en $\overrightarrow{MP}$.
La translation de vecteur $\overrightarrow{OC}$ transforme $\overrightarrow{AB}$ en $\overrightarrow{NP}$.
Les triangles M, N, P et B, C, A sont donc égaux.
Donc le rayon R du cercle circonscrit à B, C, A est égal au rayon du cercle circonscrit à M, N, P.
On va aussi montrer que le centre I de ce cercle est tel que $\overrightarrow{AI} = \overrightarrow{OP}$ et que

c'est aussi l'orthocentre du triangle ABC.



On a en effet :
Soit $\overrightarrow{AI} = \overrightarrow{OP}$
On a $\overrightarrow{PI} = \overrightarrow{OA} = \overrightarrow{CN}$ et $PC = CN$ donc
$PINC$ est un losange et donc IC est la médiatrice de PN.
de même $\overrightarrow{PI} = \overrightarrow{OA} = \overrightarrow{BM}$ et $PB = BM$ donc
$PIMB$ est un losange et donc IB est la médiatrice de PM.
Donc I est le point de concours des médiatrice de MNP c'est donc le centre du cercle circonscrit à MNP.
Puisque OP est la médiatrice de BC donc AI est la médiatrice de MN (car la translation de vecteur $\overrightarrow{0A}$ transforme B en M, C en N P en I et O en A).
Donc la médiatrice de MN passe par A puisque $MA = AN$ et elle est perpendiculaire à BC puisque BC et MN sont parallèles. donc la médiatrice de MN est la hauteur issue de A du triangle ABC.
De même la médiatrice de MP passe par B puisque $MB = BP$ et elle est perpendiculaire à AC puisque AC et MP sont parallèles. donc la médiatrice de MP est la hauteur issue de B du triangle ABC.
donc le centre du cercle circonscrit à MNP est l'orthocenntre du triangle ABC.

Démonstration avec `Xcas`
On tape :

```
O:=point(0);
U:=cercle(O,1):;U;
supposons(a=[0.3,-5,5,0.1]);
A:=point(cos(a)+(i)*sin(a));
supposons(b=[2.4,-5,5,0.1]);
B:=point(cos(b)+i*sin(b));
C1:=cercle(A,1):;C1;
```

```
C2:=cercle(B,1):;C2;
M:=normal(symetrie(droite(A,B),O));
supposons(c=[-1.6,-5,5,0.1]);
C:=point(cos(c)+(i)*sin(c));
C3:=cercle(C,1):;C3;
N:=normal(symetrie(droite(A,C),O));
P:=normal(symetrie(droite(B,C),O));
U1:=circonscrit(M,N,P):;
affichage(U1,1);
affichage(circonscrit(A,B,C),2);
I:=orthocentre(A,B,C);
U2:=cercle(I,1):; affichage(U2,4);
```

`Xcas` peut prouver que le cercle `U1` est le cercle `U2` de centre `I` (orthocentre de `ABC` ou point vérifiant $\overrightarrow{AI} = \overrightarrow{OP}$) et de rayon 1 car tous les calculs sont faits en utilisant les paramètre formels `a,b,c`.
On tape :
`tsimplify(centre(U1)-I)`
On obtient :
`0`
On tape :
`J:=translation(P-O,A)`
`tsimplify(centre(U1)-J)`
On obtient :
`0`
On tape :
`tsimplify(rayon(U1))`
On obtient :
`1` **Remarque**
On remarque que les calculs sont longs :
pour le cercle `U1:=circonscrit(M,N,P):;` (Evaluation time : 2.74)
pour `tsimplify(centre(U1)-I)` (Evaluation time : 0.54)
pour `ttsimplify(rayon(U1))` (Evaluation time : 1.12)
On peut amĺioer le temps de calcul ! ! ! ! En effet lorsque vous voulez faire faire une démonstration géométrique par `Xcas`, il est important de réduire au maximum les paramètres formels sans perte de généralités bien sûr !
Par exemple, ici, on peut supposer que `a` vaut 1 et donc que `A` se trouve sur `U` et sur l'axe des x (cette disposition n'est pas un cas particulier car on a le choix du repère). On tape :

```
O:=point(0);
U:=cercle(O,1):;U;
A:=point(1);
supposons(b=[2.4,-5,5,0.1]);
B:=point(cos(b)+i*sin(b));
C1:=cercle(A,1):;C1;
C2:=cercle(B,1):;C2;
M:=normal(symetrie(droite(A,B),O));
supposons(c=[-1.6,-5,5,0.1]);
```

```
C:=point(cos(c)+(i)*sin(c));
C3:=cercle(C,1):;C3;
N:=normal(symetrie(droite(A,C),O));
P:=normal(symetrie(droite(B,C),O));
U1:=circonscrit(M,N,P):;
affichage(U1,1);
affichage(circonscrit(A,B,C),2);
I:=orthocentre(A,B,C);
U2:=cercle(I,1):; affichage(U2,4);
```

On a alors :
pour le cercle `U1:=circonscrit(M,N,P):;` (Evaluation time : 0.92)
et ensuite `tsimplify(centre(U1)-I)` et `ttsimplify(rayon(U1))` sont instantanés.

9.3 Une suite de symétrie

On se donne trois directions d_1, d_2, d_3 et un cercle C de centre O et de rayon R et un point M_0 sur ce cercle.
On consid'ere la suite des points : M_1 est le point de C tel que M_0M_1 a pour direction d_1,
M_2 est le point de C tel que M_1M_2 a pour direction d_2,
M_3 est le point de C tel que M_2M_3 a pour direction d_3,
M_4 est le point de C tel que M_3M_4 a pour direction d_1,
M_5 est le point de C tel que M_4M_5 a pour direction d_2,
M_6 est le point de C tel que M_5M_6 a pour direction d_3,
Montrer que $M_6 = M_0$.



Démonstration

Soient $\Delta_1, \Delta_2, \Delta_3$ 3 droites passant par O et ayant comme directions, les directions perpendiculaires à d_1, d_2, d_3. Soient $S-1, S_2, S_3$ les 3 symétries droites d'axe $\Delta_1, \Delta_2, \Delta_3$.
On a :
$M_1 = S_1(M_0)$,
$M_2 = S_2(M_1)$,
$M_3 = S_3(M_2)$,
$M_4 = S_1(M_3)$,
$M_5 = S_2(M_4)$,
$M_6 = S_3(M_5) = S_3 \circ S_2 \circ S_1 \circ S_3 \circ S_2 \circ S_1(M_0)$
on sait que :
le produit des 2 symétries $S_2 \circ S_1$ est une rotation de centre 0 et d'angle $2(\Delta_1, \Delta_2)$.
le produit des 2 symétries $S_1 \circ S_3$ est une rotation de centre 0 et d'angle $2(\Delta_3, \Delta_1)$.
le produit des 2 symétries $S_3 \circ S_2$ est une rotation de centre 0 et d'angle $2(\Delta_2, \Delta_3)$.
Donc $S_3 \circ S_2 \circ S_1 \circ S_3 \circ S_2 \circ S_1$ est une rotation de centre O et d'angle :
$2(\Delta_1, \Delta_2) + 2(\Delta_3, \Delta_1) + 2(\Delta_2, \Delta_3) = 0 \bmod 2\pi$.
Donc $S_3 \circ S_2 \circ S_1 \circ S_3 \circ S_2 \circ S_1$ est l'identité.

Démonstration avec `Xcas`
On tape :

```
supposons(a=[0.6,-5,5,0.1]);
supposons(b=[1.2,-5,5,0.1]);
supposons(c=[2.0,-5,5,0.1]);
d1:=droite(y=2+a*(x+2), affichage=1);
d2:=droite(y=2+b*(x+2), affichage=2);
d3:=droite(y=2+c*(x+2), affichage=4);
C:=cercle(0,1):;C;
supposons(d=[0.6,-5,5,0.1]);
M0:=point(exp((i)*d));
M1:=symetrie(droite(y=-x/a), M0);
M2:=symetrie(droite(y=-x/b), M1);
M3:=symetrie(droite(y=-x/c), M2);
M4:=symetrie(droite(y=-x/a), M3);
M5:=symetrie(droite(y=-x/b), M4);
M6:=symetrie(droite(y=-x/c), M5);
segment(M0,M1,affichage=1);
segment(M2,M1,affichage=2);
segment(M2,M3,affichage=4);
segment(M3,M4,affichage=1);
segment(M4,M5,affichage=2);
segment(M5,M6,affichage=4);
```

On obtient :



On tape :
`affixe(M0)==normal(affixe(M6))`
On obtient :
`1`

9.4 Une suite de projections

On considère un triangle équilatèral ABC et un point M_1 sur AB.
M_1 se projette orthogonalement en H_1 sur BC,
H_1 se projette orthogonalement en K_1 sur AC et
K_1 se projette orthogonalement en M_2 sur AB etc....
On obtient ainsi sur AB une suite de points M_n.
On pose $AM_n = x_n AB$.
Calculer x_n et étudier la suite x.
On commence par faire la figure.
On écrit pour cela la suite d'instructions dans un niveau de géométrie :

```
A:=point(0);
B:=point(1);
C:=point(1/2+sqrt(3)*i/2);
triangle(A,B,C);
assume(a=[0.1,0,1]);
M:=element(segment(A,B),a);
L:=[normal(affixe(M))];
for (k:=1;k<=30;k:=k+3) {
L:=append(L,normal(affixe(projection(segment(B,C),L[k-1]))));
L:=append(L,normal(affixe(projection(segment(A,C),L[k]))));
```

```
L:=append(L,normal(affixe(projection(segment(B,A),L[k+1]))));
};
polygone_ouvert(L);
```

On obtient la figure dans l'écran de géométrie.
On rappelle que :
`M:=element(segment(A,B),a)` signifie que :
$\overline{AM} = a\overline{AB}$ et $0 \leq b \leq 1$. On rappelle aussi que :
`assume(a=[0.1,0,1])` signifie que :
la figure se fera avec `a=0.1` mais que les calculs se feront avec le paramètre formel `a` compris entre 0 et 1. On régle la fenêtre graphique :
`xyztrange(-0.1,2.0,-0.1,1.0,-10.0,10.0,-1.0,6.0,-0.1,2.0,`
`-0.146865136298,1.0,1,0.0,1.0)`
On tape L dans une entrée de commande et on pbtient :
`[a,((-i)*sqrt(3)+1)/4*a+((i)*sqrt(3)+3)/4,`
`((-i)*sqrt(3)-1)/8*a+((3*i)*sqrt(3)+3)/8,-a/8+3/8,....]`
ce qui signifie que :
$\mathtt{x}_1 = \mathsf{a}$ et $\mathtt{x}_2 = -\mathsf{a}/8 + 3/8$
Pour avoir la suite x_n on tape :
`Xn:=seq(L[k],k,0,30,3)` On trouve : $x_{11} = a/1073741824$+357913941/1073741824
On tape :
`evalf(357913941/1073741824)`
On obtient :
`0.333333333023`
Il semble donc que cette suite converge vers N tel que $AN = AB/3$.

9.4.1 La démonstration

On calcule x_2 en fonction de x_1 : on a $AM_1 = x_1 AB$
$BH_1 = (1 - x_1)/2BC$ et $CK_1 = (1 - (1 - x_1)/2)/2CA = (1 + x_1)/4CA$
$AM_2 = (1 - (1 + x_1)/4)/2AB = (3 - x_1)/8AB$
La relation de récurrence est :
$x_{n+1} = (3 - x_n)/8$ On cherche la limite l possible :
$l = (3 - l)/8$ donc $8l = 3 - l$ soit $l = 3/9 = 1/3$
La suite $u_n = x_n - 1/3$ est une suite géométrique de raison $-1/8$ puisque $u_{n+1} = x_{n+1} - 1/3 = (3 - x_n)/8 - (3 - l)/8 = -u_n/8$.
La suite $u_n = x_n - 1/3$ converge vers 0 donc la suite x_n converge vers $1/3$

9.5 Un tableau fait avec des sinusoides

On veut dessiner sur un même graphique la fonction dérivable qui vaut pour $a < b < c < d$: $\sin(x)$ sur $[a, b]$,$\sin(2 * x + \alpha)$ sur $[b, c]$ et $\sin(x)$t sur $[c, d]$ en raccordant les graphes de $\sin(x)$ et de $\sin(2 * x + \alpha)$ en des points où ils ont une même tangente horizontale. Il faut donc choisir correctement b, c, α.
On tape par exemple :

```
sinusoide0():={
  local L1,L2,L3,R,k;
  L1:=plotfunc(sin(x),x=-9*pi/2..-pi/2,affichage=epaisseur_ligne_3);
```

```
  L2:=plotfunc(sin(2*x+pi/2),x=-pi/2..3*pi/2,affichage=epaisseur_ligne_3);
  L3:=plotfunc(sin(x),x=3*pi/2..9*pi/2,affichage=epaisseur_ligne_3);
  R:=L1,L2,L3;
  retourne R;
}:;
```

On tape : `sinusoide0()`
On obtient :



On veut tracer des translatés de ces graphes selon des vecteurs de direction Oy.
On tape par exemple pour effectuer 7 translations sur la première sinusoide :

```
sinusoide1():={
  local L,R,k;
  L:=plotfunc(sin(x),x=-9*pi/2..-pi/2,affichage=epaisseur_ligne_3);
  R:=(L+k*i)$(k=-3..3);
  retourne R;
}:;
```

On tape : `sinusoide1()`
On obtient :



On tape par exemple pour effectuer 46 translations sur les 3 sinusoides :

```
sinusoide():={
  local L1,L2,L3,R,k;
  L1:=plotfunc(sin(x),x=-9*pi/2..-pi/2,affichage=epaisseur_ligne_3);
  L2:=plotfunc(sin(2*x+pi/2),x=-pi/2..3*pi/2,affichage=epaisseur_ligne_3);
  L3:=plotfunc(sin(x),x=3*pi/2..9*pi/2,affichage=epaisseur_ligne_3);
  R:=(L1+k/4*i)$(k=-20..25),(L2+k/4*i)$(k=-20..25),(L3+k/4*i)$(k=-20..25);
```

```
  retourne R;
}:;
```

On tape : `sinusoide()`
On obtient une sinusoide.

9.6 La suite de Syracuse

Soit a un entier positif. On veut étudier avec des graphiques la suite de Syracuse définie par :
$u_0 = a$
$u_n = u_{n-1}/2$ si u_{n-1} est pair et
$u_n = 3 * u_{n-1} + 1$ si u_{n-1} est impair.
Cette suite se termine toujours (?) par 1,4,2,1,4,2,1... mais on ne sait pas le montrer.
Pour étudier cette suite on peut :
- utiliser le tableur en mettant dans `A0` la valeur a de départ et dans `A1` la formule :
`=ifte(irem(A0,2)==0,iquo(A0,2),3*A0+1)` ou encore
`=if ((irem(A0,2))==0) iquo(A0,2); else 1+3*A0;`
- utiliser un programme `syracuse` qui renvoie le maximum de cette suite et le nombre d'éléments de cette suite et `syracuse0` qui écrit en plus les termes de la suite ou encore `syracuse100` qui renvoie le maximum, le nombre de termes et la valeur de départ de la plus longue suite démarrant par un nombre entre 2 et 100.

```
  syracuse(a):={
    local k,m;
    k:=0;
    m:=a;
    while (a!=1) {
      if (irem(a,2)==0) a:=iquo(a,2);
      else {
a:=a*3+1;
if (a>m){m:=a};
      }
      k:=k+1;
    }
  retourne m,k;
}:;

 syracuse0(a):={
    local m,k;
    m:=a;
    k:=0;
    print(a);
    while (irem(a,2)==0){
      a:=iquo(a,2);
      k:=k+1;
      print(a);
    }
```

```
    while(a!=1){
      a:=3*a+1;
      k:=k+1;
      print(a);
      if (m<a) {m:=a;}
      while (irem(a,2)==0){
a:=iquo(a,2);
k:=k+1;
    print(a);
      }
    }
    return(m,k);
  };

  syracuse100():={
    local k,kn,kt,l,lt,m,mt;
    lt:=0;
    for (k:=2;k<101;k:=k+1){
      kn:=k;
      m:=k;
      l:=0;
      while (kn!=1) {
if (irem(kn,2)==0) kn:=iquo(kn,2);
else {
  kn:=kn*3+1;
  if (m<kn) {
    m:=kn;
  }
}
l:=l+1;
      }
      if (l>lt) {
mt:=m;lt:=l;kt:=k;
      }
    }
    return(mt,lt,kt);
  };
```

On ouvre un éditeur de programme, on recopie la procédure, puis grâce au bouton `OK` le programme est validé.
On tape `syracuse100()`, on trouve :
`9232,118,97` ce qui veut dire que c'est en démarrant avec 97 que la suite a le plus de termes (ici 118 termes) et le maximum de cette suite est 9232.
On peut bien sûr modifier les paramètres de la boucle `for` en mettant par exemple :
`for (k:=101;k<200;k:=k+1)`
On tape `syracuse100()`, on trouve alors :
`250504,124,177`
ou encore :
`for (k:=901;k<1000;k:=k+1)`

On tape `syracuse100()`, on trouve alors :
`250504,173,937`
On peut encore modifier facilement pour savoir si un plus grand nombre de terme donne la plus grande valeur atteinte (cela semble vrai ! ! !) en changeant pour cela :
`if (l>lt) {mt:=m;lt:=l;kt:=k;}` en
`if (m>mt) {mt:=m;lt:=l;kt:=k;}` et rajouter au début `kt:=2;`.
- utiliser un programme qui dessine les points (n, u_n) lorsqu'on donne en entrée $u_0 = a$
- lorsqu'on donne en entrée $u_0 = a$ et en notant n la première valeur de k pour laquelle $u_k = 1$ et m le maximum des u_k pour $k \leq n$, dessiner les points a, m ou encore dessiner les points a, u_k pour $k = 0..n$.
On écrit la procédure `syracuse1` (resp `syracuse2`) qui dessine les points (k, u_k) (resp les segments reliant les points (k, u_k)) dans l'écran de géométrie et la procédure `syracuse3` qui dessine les points a, u_k pour k allant de 0 à n et cela pour a allant de 2 à 100 :

```
  syracuse1(a):={
    local m,k;
    m:=a;
    k:=0;
    point(0,a);
    while (irem(a,2)==0){
      a:=iquo(a,2);
      k:=k+1;
      point(k,a);
    }
    while(a!=1){
      a:=3*a+1;
      k:=k+1;
      point(k,a);
      if (m<a) {m:=a;}
      while (irem(a,2)==0){
a:=iquo(a,2);
k:=k+1;
point(k,a);
      }
    }
    return(m,k);
  };


  syracuse2(a):={
    local m,k,k0,a0;
    m:=a;
    k:=0;
    point(k,a);
    while (irem(a,2)==0){
      k0:=k;
      a0:=a;
```

```
      a:=iquo(a,2);
      k:=k+1;
      segment(k0+i*a0,k+i*a);
    }
    while(a!=1){
      k0:=k;
      a0:=a;
      a:=3*a+1;
      k:=k+1;
      segment(k0+i*a0,k+i*a);
      if (m<a) {m:=a;}
      while (irem(a,2)==0){
k0:=k;
a0:=a;
a:=iquo(a,2);
k:=k+1;
segment(k0+i*a0,k+i*a);
      }
    }
    return(m,k);
  };

  syracuse3():={
    local k,kn;
    for(k:=2;k<101;k:=k+1){
      point(k,k);
      kn:=k;
      while (kn!=1) {
if (irem(kn,2)==0) kn:=iquo(kn,2); else kn:=kn*3+1;
point(k,kn);
      }
    }
  };
```

Ne pas oubler de régler la fenêtre graphique en mettant par exemple : `X-=Y-=WX-=WY-=0`, `X+=WX+=100` et `Y+=WY+=1000`.
puis on tape par exemple `syracuse1(123)`.

9.7 La suite des tas

On dispose k jetons en p tas.
On construit la suite des tas de la façon suivante :
on prend un jeton dans chaque tas et tous ces jetons forment un nouveau tas qui sera le dernier tas, et on recommence.
Ce qui est sûr c'est que cette suite de tas est pérodique puisque il n'y a qu'un nombre fini de façons de disposer k jetons en tas.
Il s'agit de voir comment se comporte cette suite.
On peut montre que lorsque $k = n * (n + 1)/2$ cette suite stationne en :

$1, 2, 3, ..., n$.

9.7.1 Une remarque

Lemme La suite débutant par $1, 2, 3, ..., n$ non ordonnè stationne vers $1, 2, 3, ..., n$. En effet, si on répartit les jetons en n tas $1, 2, 3, ..., n$ de façon non ordonnée, on aura la suite ordonnée $1, 2, 3, ..., n$ au bout d'au plus $n - 1$ manipulations.
En effet à la première étape n se trouvera en dernier, à la deuxième étape $n - 1, n$ se trouveront à la fin, et à la $(n-1)$-ième étape $2, ...n-1, n$ se trouvera en dernier et on aura donc obtenu $1, 2...n$ puisque le nombre k de jetons vaut $n(n - 1)/2$.

9.7.2 Le programme de simulation

```
//programme de simulation tas.cxx
tas(l):={
local s,j,k,lr;
lr:=[l];
while (1) {
s:=size(l);
for (j:=0;j<s;j++) {
l[j]:=l[j]-1;
}
l:=concat(l,s);
//on supprime les zeros de l
k:=0;
for (j:=0;j<s+1;j++){
if (l[j]!=0){
l[k]:=l[j];
k:=k+1;
}
}
l:=mid(l,0,k);
if (member(l,lr)) return lr;
lr:=append(lr,l);
}
}
```

On tape :
`tas([10])`
On obtient :
```
[[10],[9,1],[8,2],[7,1,2],[6,1,3],[5,2,3],
[4,1,2,3],[3,1,2,4],[2,1,3,4],[1,2,3,4]]
```

Chapitre 10

Pour s'amuser avec les mesures

10.1 La ficelle et la terre

Imaginez qu'avec une ficelle vous fassiez le tour de la terre. Puis vous rajoutez un mètre à cette ficelle et vous ceinturez à nouveau la terre. À quelle distance du sol va alors se trouver la ficelle ?
Même question avec une balle de tennis.

Si r est le rayon de la terre en mètre (ou de la balle de tennis), son périmètre est donc : $2\pi r$. La ficelle va donc mesurer $2\pi r + 1$ et cela correspond à un rayon R vérifiant $2\pi r + 1 = 2\pi R$. Donc $2\pi(R - r) = 1$ c'est à dire $R - r = 1/(2\pi)$.
On tape :
```
evalf(1/2/pi)
```
On obtient :
```
0.159154943092
```
Donc quelque soit le rayon de la sphère la ficelle va être à environ 15.9 cm de la surface de la sphère.

10.2 Les lunules d'Hippocrate

Ce théorème, très ancien, a été démontré par Hippocrate de Chios (-500) (Ne pas le confondre avec Hippocrate de Cos, le médecin), qui étudia aussi la duplication du cube, c'est-à-dire le calcul de la racine cubique de 2.
Hippocrate recherchait alors la quadrature du cercle et pensait que la quadrature de ses lunules allait le rapprocher du but.
Une lunule est une portion de surface délimitée par deux arcs de cercles non concentriques de rayons différents. Ces arcs ont mêmes extrémités et forment un croissant de lune en forme de ménisque : convexe d'un côté et concave de l'autre.
Construction
Soit le triangle ABC rectangle en A et $\mathcal{C}$ le cercle circonscrit à ABC (de diamètre BC).
La lunule L_{AC} est la figure formée par le demi-disque de diamètre AC extérieur au triangle ABC, auquel on enlève son intersection avec le disque délimité par $\mathcal{C}$.
La lunule L_{AB} est la figure formée par le demi-disque de diamètre AB extérieur au triangle ABC, auquel on enlève son intersection avec le disque délimité par $\mathcal{C}$.

Ces deux lunules (en jaune sur la figure ci-dessous) sont appelées **lunules d'Hippocrate**.
La somme des aires de ces deux lunules L_{BC} et de L_{BA} est égale à l'aire du triangle



ABC (en rouge).
On calcule $S1$ l'aire du triangle ABC :
$S1 = AB * AC/2$ On calcule $S2$ la somme des aires de L_{BC} et de L_{BA} :
$S2$ est obtenue par différence : $S2$ est égale à la somme des aires des demi-cercles de diamètres AB et AC à laquelle on enlève l'aire en bleu.
L'aire en bleu est égale à l'aire du demi-cercle de diamètre BC à laquelle on enlève l'aire $S1$ du triangle ABC :
$S2 = \pi * AB^2/2 + \pi * AC^2/2 - (\pi * BC^2/2 - S1)$
D'après le théorème de Pythagore on a : $AB^2 + AC^2 = BC^2$ donc :
$S2 = S1$.
Pour faire la figure on a tapé :

```
A:=point(0);
B:=point(2,affichage=quadrant1);
C:=point(3*i,affichage=quadrant1);
cercle(A,C,0,pi,affichage=3+rempli);
cercle(A,B,-pi,0,affichage=3+rempli);
cercle(B,C,0,pi,affichage=4+rempli);
triangle(A,B,C,affichage=1+rempli);
```

`Xcas` sait remplir de couleur les polygones et les secteurs circulaires. On est donc obligé de procéder par superposition de couleur pour remplir les lunules d'Hippocrate et cela symbolise les opérations que l'on fait pour calculer l'aire des 2 lunules.
Exercice
Soient un carré de côtés l et les cercles ayant comme diamètre les côtés du carré.
À l'extérieur du carré, ces cercles déterminent avec le cercle circonscrit au carré 4 lunules.

Trouver l'aire des 4 lunules ainsi déterminées.
À l'intérieur du carré, ces cercles en se coupant déterminent 4 "pétales".
Trouver l'aire des 4 "pétales" ainsi déterminées.
Pour faire la figure, on tape dans un niveau de géométrie :

```
cercle(point(2),2,-pi/2,pi/2,affichage=1+rempli);
cercle(point(2*i),2,0,pi,affichage=2+rempli);
cercle(point(-2),2,pi/2,3*pi/2,affichage=3+rempli);
cercle(point(0,-2),2,pi,2*pi,affichage=4+rempli);
cercle(0,2*sqrt(2),affichage=5+rempli);
cercle(point(0,-2),2);
cercle(point(0,2),2);
cercle(point(2,0),2);
cercle(point(-2,0),2);
carre(-2-2*i,2-2*i);
```



La solution Un carré est formé de 2 triangles rectangles donc l'aire des 4 lunules est égale à l'aire du carré (cl le résultat précédent).
La somme de l'aire du carré et de l'aire des "pétales" est égale à l'aire des 4 demi-cercles de rayon $l/2$ (car les 4 demi-cercles qui sont à l'intérieur du carré remplissent le carré et se superposent selon les "pétales") donc l'aire des "pétales" vaut :

$$\frac{\pi}{2}l^2 - l^2$$

.

10.3 D'autres résultats obtenus par Hippocrate

Si la tentative de la quadrature du cercle est un échec, Hippocrate a trouvé la quadrature de plusieurs lunules en partant de la remarque suivante :
deux secteurs circulaires OAB et $O_1A_1B_1$ de rayon R et R_1 semblables i.e. dont

les angles au centre t ont la même valeur ont des aires proportionnelles aux carrés des longueurs de leurs cordes.
Chaque secteur circulaire est composé d'un triangle et d'une calotte circulaire. Les aires des triangles ainsi que les aires des calottes circulaires sont aussi proportionnelles aux carrés des longueurs des cordes.



Dans la figure ci-dessus on a :
`longueur2(A1,B1)/longueur2(A,B)` renvoie `16/9`.
Cela signifie donc que :
$16S = 9S1$ où S désigne l'aire de la calotte verte et S_1 désigne l'aire de la calotte jaune, i.e. $S1 = 16S/9$.
Si deux secteurs circulaires OAB et $O_1A_1B_1$ de rayon R et R_1 ont pour angle au centre t on a (dans la figure $R = 3$, $R1 = 4$ et $t = 5\pi/12$) :
La longueur AB vaut $2R\sin(t/2$ donc $R = AB/(2\sin(t/2)$
L'aire du secteur circulaire vaut $tR^2/2 = AB^2(t/(8\sin(t/2)^2)$
L'aire du triangle OAB vaut $R^2\cos(t/2)\sin(t/2) = AB^2(1/(4\tan(t/2))$
L'aire de la calotte circulaire vaut $AB^2((t/(8\sin(t/2)^2) - (1/(4\tan(t/2))$.
Pour faire la figure d'un secteur circulaire OAB, de centre O, de rayon r, d'angle au centre a radians et tel que l'angle $(Ox, OA) = t$, on tape :

```
fonction Secteurcirc(O,r,a,t,ct,cc)
local A,B,L,xO,yO;
L:=NULL;
[xO,yO]:=coordonnees(O);
A:=point(xO+r*cos(a),yO+r*sin(a));
B:=point(xO+r*cos(t+a),yO+r*sin(t+a));
L:=L,affichage(cercle(O,r,a,t+a),cc+rempli);
L:=L,triangle(O,A,B,affichage=ct+rempli);
retourne L;
ffonction:;
```

10.3.1 L'aire d'une lunule égale à l'aire d'un triangle isocèle

Soit un triangle ABC isocèle et rectangle en C.
Soit D le symétrique de C par rapport à AB.
Le demi-cercle ABC de diamètre AB et le quart de cercle AMB de centre D définissent la lunule $AMBC$ en rouge sur la figure.



On a $AB^=2*CB^2$ d'après Pythagore puisque ABC est rectangle en C.
Grâce à Hippocrate, on va montrer que l'aire de cette lunule est égale à l'aire du triangle ABC (en bleu sur la figure obtenue par translation de $ABNC$) :
$AB^2 = 2*CB^2$, donc l'aire de la calotte verte ABM est égale à 2 fois l'aire de la calotte BNC donc l'aire verte est égale à l'aire magenta (cf la figure obtenue par translation de $ABNC$).
On a donc l'aire rouge est égale à l'aire bleue puisque :
aire($AMBNC$)=aire(ABC)-aire(ABM)+2*aire(BNC)=aire(ABC).

10.3.2 L'aire d'une lunule égale à l'aire d'un trapèze isocèle

Soit un trapèze isocèle $ABCD$ vérifiant $BC = CD = AD$ et $AB^2 = 3DC^2$.
Soit l'arc de cercle $BCMDA$ de centre I intersection des médiatrices de AC et de DC (où M est un point de l'arc CD).
Soit l'arc de cercle BNA tangent à AC de centre J intersection de la médiatrice de DC et de la perpendiculaire en A à AC.



L'aire de la lunule rouge $ANBCMD$ est égale à l'aire du trapèze $ABCD$.
En effet, l'aire de la calotte bleue est égale à 3 fois l'aire de la calotte magenta (cf la figure obtenue par translation de $ABCMD$) car les secteurs circulaires $JBNA$ et $ICMD$ sont homothétiques et que $AB^2 = 3DC^2$.
On a donc l'aire rouge est égale à l'aire verte puisque $AD = DC = CB$ et :
aire($ANBCMD$)=aire($ABCD$)-aire(BNA)+3*aire(CMD)=aire($ABCD$).

10.3.3 L'aire d'une lunule égale à l'aire d'un pentagone concave

Soit un trapèze isocèle $ABCD$ vérifiant $BC = CD = AD$ et soit E l'intersection de ses diagonales AC et BD.
Le cercle circonscrit au triangle ABE est tangent aux droites AD et BC.



En effet :
le centre J du cercle c circonscrit à ABE est l'intersection de la médiatrice de AE et de la médiatrice de AB. Si M est le milieu de AE on a (cf figure) :
$(\overrightarrow{JE}, \overrightarrow{JM}) = (\overrightarrow{AC}, \overrightarrow{AD}) = t$
donc $(\overrightarrow{AJ}, \overrightarrow{AD}) = \pi/2$.
donc les droites CB et DA sont tangentes à c.
La puissance de C par rapport au cercle c vaut :
$CE * CA = CB^2$ puisque la droite CB est tangente à c.
On impose maintenant que $2AE^2 = 3DC^2$.
Il faut faire maintenant une figure qui respecte l'égalité :
$2AE^2 = 3BC^2$ ou encore $AE = BC\sqrt{6}/2$.
On pose :
$BC = BD = DA = l$, donc on a :
$CB^2 = l^2 = CE * CA = (CA - AE) * CA = CA^2 - CA * l\sqrt{6}/2$
donc la longueur CA est solution de l'équation :
$x^2 - x\sqrt{6}l/2 - l^2 = 0$.
On tape (pour faire la figure on choisit $l = 2$) :
`solve(x^2-x*sqrt(6)-4=0)`
On obtient :
`[(sqrt(6)-(sqrt(22)))/2,(sqrt(6)+sqrt(22))/2]`
la longueur CA est positive donc $CA = (\sqrt{6} + \sqrt{22}))/2$.
On trace le segment DC, puis on définit A comme intersection du cercle $c1$ de centre D et de rayon 2 et du cercle $c2$ de centre C et de rayon $(\sqrt{6} + \sqrt{22}))/2$ en choisissant A en dessous de DC.
On tape :

```
C:=point(1,affichage=quadrant1);
D:=point(-1,affichage=quadrant2));
c1:=cercle(D,2):;
c2:=cercle(C,(sqrt(6)+sqrt(22))/2):;
```

```
A:=inter_unique(c1,c2,point(-2*i));
B:=symetrie(droite(x=0),A);
polygone(A,B,C,D);
E:=inter_unique(droite(A,C),droite(x=0));
segment(A,C);
segment(B,D);
J:=inter_unique(perpendiculaire(A,droite(A,D)),droite(x=0));
I:=inter_unique(mediatrice(A,D),droite(x=0));
cercle(I,longueur(I,A),angle(I,I+2,B),angle(I,I+2,A),affichage=4+rempl
polygone(A,B,C,D,affichage=3+rempli);
cercle(J,longueur(A,J),angle(J,J+2,B), angle(J,J+2,A),affichage=1+remp
triangle(J,A,E,affichage=2+rempli);
triangle(J,B,E,affichage=2+rempli);
A:=affichage(A,quadrant2);
D:=affichage(D,quadrant2);
E:=E;
polygone(A,B,C,D);
J:=affichage(J,quadrant1);
I:=I;
M:=milieu(A,D);
N:=milieu(D,C)
segment(M,I);
segment(N,J);
segment(I,D);
segment(I,C);
angle(J,E,A,"t");
angle(I,C,D,"t");
angle(I,N,M,"t");
```

On obtient :



Les 2 aires rouges sont égales aux 3 aires bleues car on a :
$AE^2 = BE^2 = 3AD^2/2 = 3BC^2/2 = 3DC^/2$ et
les secteurs circulaires JEA et IDA sont semblables puisque :
$(\overrightarrow{JE}, \overrightarrow{JA}) = (\overrightarrow{IC}, \overrightarrow{ID})$ (cf figure).
L'aire de la lunule $AEBCD$ (aire jaune +aire bleue) est donc égale à l'aire du pentagone concave $AEBCD$ (aire jaune +aire rouge).

10.3.4 L'aire d'une lunule et d'un cercle égale à l'aire d'un triangle et d'un hexagone

Soit un cercle c_1 de centre O et de rayon r et soit $ABCDEF$ un hexagone inscrit dans ce cercle.
Soit un cercle c_2 de centre O et de rayon $R = \sqrt{6}r$.
Les droites OA, OB et OC coupent c_2 respectivement en G, H et I.
Soit c_3 le cercle tangent en G à HG et tangent en I à HI. OH coupe c_3 en N.
Hippocrate montre que l'aire de la lunule $INGH$ plus l'aire du cercle c_1 est égale à l'aire du triangle GHI plus l'aire de l'hexagone.



Les secteurs circulaires JGH, OGH et OAB sont semblables car :
$(\overrightarrow{OG}, \overrightarrow{OH}) = (\overrightarrow{JG}, \overrightarrow{JI} = \pi/3)$.
On a $IG^2 = (\sqrt{3}R)^2 = 3R^2 = 18r^2$, $GH^2 = R^2 = 6r^2 = IG^2/3$.
Soit S_1 l'aire de la calotte magenta AB, S_2 l'aire de la calotte rouge GH, S_3 l'aire de la calotte jaune GI.
On a d'après la remarque d'Hippocrate $S_3 = 3S_2 = 18S_1$ et donc :
l'aire magenta vaut $6S_1 = S_2$, l'aire jaune vaut $3S_2$,l'aire rouge vaut $2S_2$.
Donc :
aire lunule$INGH$=aire rouge+aire verte=aire rouge+aire triangleGHI-aire jaune.
aire lunule$INGH$=$2S_2 - 3S_2$+aire triangleGHI=aire triangleGHI-S_2
aire c_1=aire magenta+aire hexagone=aire hexagone+S_2
donc en ajoutant membre à membre on a :
aire lunule$INGH$+aire c_1=aire triangleGHI+aire hexagone.

10.4 Le théorème de Pappus d'Alexandrie (300 ap.J.-C.)

Le théorème de Pappus d'Alexandrie est aussi une généralisation du théorème de Pythagore.
On construit à l'extérieur d'un triangle ABC trois parallélogrammes :

— un parallélogramme $ABEG$ quelconque
— un parallélogramme $ACFD$ quelconque
— Soit H l'intersection de EG et DF. Le parallélogramme $BJKC$ est tel que $\overrightarrow{BJ} = \overrightarrow{HA}$



Alors

$$\texttt{Aire}(BJKC)\texttt{=Aire}(ACFD)\texttt{+Aire}(ABEG)$$

On va montrer que cette égalité se visualise facilement car les surfaces rouges ont même aire et les surfaces vertes ont aussi même aire.
Soient L l'intersection de HA et de JK, 0 l'intersection de HA et de BC, M l'intersection de BJ et de EG et N l'intersection de DF et de CK.
On a :
Aire($ABGE$)=Aire($ABMH$) parallélogrammes ayant même base AB et même hauteur
Aire($ABMH$)=Aire($BJLO$) parallélogrammes ayant une base égale ($HA = BJ = OL$) et même hauteur
donc les deux surfaces rouges ont même aire.
La même démonstration est valable pour les deux surfaces vertes. Si on construit à l'extérieur d'un triangle ABC deux carrés et un parallélogramme :

— un carré $ABGE$,
— un carré $ACDF$,
— H l'intersection de EG et DF.
— le parallélogramme $BJKC$ tel que $\overrightarrow{BJ} = \overrightarrow{HA}$,
— un carré $CBPQ$.
— R l'intersection de BP et JK.

— S l'intersection de CQ et JK.

Le carré $CBPQ$ a pour aire $BC^2 = a^2$ Le parallélogramme $BJKC$ a pour aire $AB^2 + AC^2 = c^2 + b^2$ On sait que $a^2 = b^2 + c^2 - 2bc\cos(A)$ Donc l'aire du parallélogramme $BJKC$ est égale à l'aire du carré $CBPQ$ plus $2 * b * c\cos(A)$.
L'aire du parallélogramme $BJKC$ est égale à l'aire du rectangle $BCRS$
Donc $2 * b * c\cos(A)$ est l'aire du rectangle $QPRS$



10.5 Aire d'un cercle troué

On perce un cercle de rayon R avec un cercle de même centre et de rayon r. Quelle est l'aire de ce cercle troué. Exprimer cette aire en fonction de $d = \sqrt{R^2 - r^2}$.
On sait que l'aire d'un cercle de rayon R est : πR^2.
L'aire du cercle troué est donc : $\pi(R^2 - r^2) = \pi d^2$ L'aire d'un cercle troué est égale à l'aire du cercle de rayon d où d est la longueur de la demi-corde qui est tangente au trou.

10.6 Volume de la sphère trouée

10.6.1 Volume de la calotte de hauteur h d'une sphère de rayon R

On note R le rayon de la spère, r le rayon de la base de la calotte et h la hauteur de la calotte et on suppose que $0 < h < R$. 0n a donc : $r^2 = R^2 - (R-h)^2$
On tape :
`assume(R>0 and h>0 and h<R)`
On calcule une intégrale triple :
$\int_0^{2*\pi}(\int_0^r(\int_0^{\sqrt{R^2-r^2}} dz) * \rho d\rho)d\theta$
On tape :

```
factor(int(int(int(1,z,R-h,sqrt(R^2-ro^2))*ro,
        ro,0,sqrt(R^2-(R-h)^2)),t,0,2*pi))
```

On obtient :
`(h^2*pi*(3*R-h))/3`
Donc le volume de la calotte de hauteur h d'une spère de rayon R est :

$$V_C = \pi h^2 \frac{3*R-h}{3}$$

Vérification On sait que le volume d'une sphère de rayon R est : $\frac{4}{3}\pi R^3$ Le volume de 2 calottes sphériques de rayon R a pour hauteur R est :
$2\pi R^2 \frac{3*R-R}{3} = \frac{4}{3}\pi R^3$

10.6.2 Volume d'une sphère trouée

On perce une sphère de rayon R avec un cylindre ayant pour axe un diamètre de la sphère et comme base un cercle de rayon r. Quelle est le volume de cette sphère trouèe. Exprimer ce volume en fonction de $d = \sqrt{R^2 - r^2}$.

Avec les notations précédentes on a enlevé à la sphère :
un cylindre ayant comme base un cercle de rayon r comme hauteur $2d$ et comme volume $2\pi r^2 d = 2\pi(R^2 - d^2)d$ et,
2 calottes sphériques ayant comme base un cercle de rayon r et ayant comme hauteur $h = R - d$.
On sait que le volume d'une sphère de rayon R est : $\frac{4}{3}\pi R^3$ Le volume de cette sphère trouèe est donc :

$$\frac{4}{3}\pi R^3 - 2\pi h^2 \frac{3*R-h}{3} - 2\pi r^2 d$$

On tape :
`factor(2*pi/3*(2*R^3-(R-d)^2*(3*R-(R-d))-3*(R^2-d^2)*d))`
On obtient :
`(4*d^3*pi)/3`
Donc le volume d'une spère de rayon R trouée par un cylindre d'axe un diamètre et de hauteur $2d$ est :

$$V_{ST} = \frac{4}{3}\pi d^3$$

c'est à dire le volume d'une spère de rayon R trouée par un cylindre d'axe un diamètre et de hauteur $2d$ est égal au volume d'une sphère de rayon d.

Chapitre 11

Les puzzles

11.1 Puzzle transformant un triangle en un triangle de base et de hauteur égales

Soient un quadrilatère $ABDC$ et I (resp J) la projection orthogonale de A (resp D) sur BC tel que $AI = DJ$.
Les triangles ABC et BCD ont donc une même base (BC) et des hauteurs égales ($AI + AJ$).
Le segment AD coupe BC en K.

— K **se trouve entre** B **et** C (c'est vrai si les angles des triangles sont aigus).
On mène par K des parallèles aux côtés du triangle situé de l'autre côté.
On tape :

```
supposons(a=[0,-5,5,0.1]);
A:=point(a+3*i);
B:=point(-2);
C:=point(4);
D:=point(-1-3*i);
polygone(A,B,D,C);
I:=point(a);
J:=point(-1);
segment(A,I,affichage=2+ligne_tiret_point);
segment(D,J,affichage=2+ligne_tiret_point);
segment(B,C);
segment(A,D);
K:=inter_unique(droite(B,C),droite(A,D));
L:=inter_unique(segment(A,C), parallele(K,droite(D,C)));
segment(K,L);
M:=inter_unique(segment(B,A), parallele(K,droite(D,B)));
segment(K,M);
N:=inter_unique(segment(B,D), parallele(K,droite(A,B)));
segment(K,N);
P:=inter_unique(segment(C,D), parallele(K,droite(A,C)));
segment(K,P);
```

On obtient (pour $a = 0$) :



K est le milieu de AD puisque $AI = DJ$.
On en déduit que :
En considérant le triangle ADC on en déduit que :
L est le mileu de AC et $KL = DP = PC$ et
P est le mileu de DC et $KP = AL = CL$
En considérant le triangle ABD on en déduit que :
M est le mileu de AB et $KM = DN = NB$ et
N est le mileu de BD et $KL = DP = PC$.
Donc le puzzle qui transforme ABC en BDC a 4 morceaux, on tape :

```
affichage(triangle(B,M,K),1+rempli);
affichage(triangle(A,M,K),2+rempli);
affichage(triangle(A,L,K),3+rempli);
affichage(triangle(C,L,K),4+rempli);
affichage(triangle(B,N,K),1+rempli);
affichage(triangle(D,N,K),2+rempli);
affichage(triangle(D,P,K),3+rempli);
affichage(triangle(C,P,K),4+rempli);
segment(B,C)
```

On obtient :



On remarque que la pièce jaune et la pièce verte peuvent se reunir.
On obtient donc un puzzle de 3 pièces :



On remarquera que `translation(A-K,polygone(K,N,D,P))` renvoie `polygone(A,M,K,L)`.

— **K est en B ou en C**
Dans l'exemple K est en B pour $a = -3$.
On obtient alors un puzzle ayant 2 pièces :



On remarquera que les angles B des 2 triangles sont supplémentaires : l'un est obtu et l'autre est aigu (sauf si les 2 triangles sont égaux !).

— **K est à l'extérieur du segment BC**
Supposons que l'angle B du triangle ABC soit obtus et que K se trouve à l'extérieur du segment BC.
On forme le parallélogramme $ABCA1$ ($A1$ est le symétrique de B par rapport au milieu L de AC.



Au triangle ABC on associe le triangle $A1BC$: ces 2 triangles ont même aire, ont le triangle BCL en commun et 2 triangles égaux ABL et $CA1L$ car symétrique par rapport à L.
On peut donc avec 2 pièces transformer le triangle ABC en le triangle $A1BC$: le triangle $A1BC$ a même base, même hauteur que le triangle ABC et $ABCA1$ est un parallélogramme.

Exemple pour $a = -5$
Les points qui nous seront utiles :

```
A:=point(a+3*i);
B:=point(-2);
C:=point(4);
D:=point(-1-3*i);
polygone(A,B,D,C);
I:=point(a);
J:=point(-1);
segment(A,I,affichage=2+ligne_tiret_point);
segment(D,J,affichage=2+ligne_tiret_point);
segment(B,C);
segment(A,D);
K:=inter_unique(droite(B,C),droite(A,D));
L:=milieu(A,C);
M:=milieu(A,B);
N:=milieu(D,B);
P:=milieu(D,C);
segment(K1,P);
A1:=symetrie(milieu(A,C),B);
M1:=symetrie(milieu(A,C),M);
K1:=inter_unique(droite(A1,D),droite(B,C));
segment(A1,D);
segment(A1,C);
segment(A1,B);
segment(K1,M1);
```

```
segment(K1,L);
segment(K1,P);
segment(K1,N);
R1:=inter_unique(segment(C,L),segment(K1,M1));
R:=translation(K1-A1,R1);
S:=symetrie(milieu(K1,C),R1);
droite(B,C);
segment(N,R);
segment(K1,S);
```



Les 2 pièces qui faut partager :L

```
affichage(triangle(B,L,K1),1+rempli);
affichage(triangle(A1,L,K1,M1),2+rempli);
affichage(triangle(C,M1,K1),4+rempli);
affichage(triangle(B,N,K1),1+rempli);
affichage(polygone(D,N,K1,P),2+rempli);
affichage(triangle(C,P,K1),4+rempli);
segment(B,C);
triangle(A,B,C);
segment(C,N);
A:=A;D:=D;B:=B;C:=C;A1:=A1;
```



On rajoute donc :

```
affichage(triangle(D,N,Q),5+rempli);
P5:=affichage(triangle(A1,L,Q1),5+rempli);
P6:=affichage(polygone(A1,Q1,R1,M1),6+rempli);
affichage(polygone(D,Q,R,P),6+rempli);
P0:=affichage(triangle(C,M1,R1),47+rempli):;P0;
affichage(triangle(C,P,R),47+rempli);
```

On voit les 5 pièces du puzzle et la transformation de ABC en DBC et en $A1BC$.



Ou encore avec 7 pièces :



11.2 Exercice

Soient les triangles :
ABC ($AB = AC = (\sqrt{6} - \sqrt{2})/2$ et $(\overrightarrow{BC}, \overrightarrow{BA}) = 5\pi/6$ et
DBA ($DB = DC = (\sqrt{3} - 1)/2$ et $(\overrightarrow{DB}, \overrightarrow{DA}) = \pi/2$.
Fabriquer un puzzle transformant le triangle ABC en le triangle ABD.
Une solution
Le triangle DBA est rectangle isocèle donc on a bien :
$AB = DB\sqrt{2} = (\sqrt{6} - \sqrt{2})/2 = AC$
L'aire de DBA vaut donc :
$(\sqrt{3} - 1)^2/8 = (2 - \sqrt{3})/4$
Le triangle ABC est isocèle, d'angles à la base de $\pi/12$ donc :
$AC = 2AB\cos(\pi/12) = (\sqrt{6} - \sqrt{2}) * (sqrt(2) + sqrt(6))/4 = 1$
et la hauteur h issue de B vaut :
$h = AB\sin(\pi/12) = (2 - \sqrt{3})/2$ L'aire de ABC est donc :
$AC * h/2 = (2 - \sqrt{3})/4$.
Les triangles ABC et DBA ont donc même aire et un côté AB en commun donc les hauteurs issues de C et de D sont égales. On est dans le cas où l'intersection de AB avec CD se trouve à l'extérieur du segment AB.



On forme donc le losange $ABCA1$.
Ainsi, ABC at $ABA1$ ont même aire (c'est la moitié de l'aire du losange $ABCA1$). ABC at $ABA1$ ont en commun le triangle ABI (L étant le milieu de AC etle milieu de $A1B$ puisque c'est l'intersection des diagonales d'un losange) et le triangle $A1LA$ est égal à $A1LC$.
Soit K l'intersection de AB avec $A1D$ (K est le milieu de $A1D$ puisque $A1$ et D sont à égals distance de la droite AB).
On peut faire un puzzle de 3 pièces pour réaliser les triangles : ABD et $ABA1$ qui sont :
AKN ou AKM ($ANKM$ est un parallélogramme)
KJB ou KBL ($KJBL$ est un parallélogramme)
$NDJK$ ou $MKNA1$ (M (resp N J) est le milieu de $AA1$ (resp AB,DB) puisque K est le milieu de $A1D$.
On trace le segment AC qui partage les pièces AKM et $MKLA1$ en deux morceaux : AKP et APM forment AKM et PKL et $MPLA1$ forment $MKLA1$.
On a alors :
APM=$CM1P1$ et $MPLA1$=$M1P1LB$.
On obtient donc 5 pièces.
On tape :

```
A:=point(sqrt(2)/4-sqrt(6)/4;
B:=point(-sqrt(2)/4+sqrt(6)/4);
D:=point(-i*(-sqrt(2)/4+sqrt(6)/4));
C:=rotation(B,-5*pi/6,A);
p1:=polygone(A,B,C):;p1;
p2:=polygone(A,B,D):;p2;
L:=milieu(A,C);
A1:=symetrie(L,B);
p3:=polygone(A,B,A1):;p3;
K:=inter_unique(droite(y=0),droite(A1,D));
J:=inter_unique(parallele(K,droite(B,A1)) ,droite(D,B));
```

```
polygone(J,B,L,K);
M:=inter_unique(parallele(K,droite(D,A)) ,droite(A,A1));
N:=inter_unique(parallele(K,droite(A,A1)) ,droite(A,D));
polygone(M,A,N,K);
M1:=milieu(B,C);
P:=inter_unique(droite(M,K),droite(A,L));
P1:=symetrie(L,P);
P2:=translation(P-M,N);
segment(P2,J);
segment(P1,M1);
P3:=symetrie(milieu(A,K),P);
segment(P3,K);
affichage(polygone(C,P1,M1),1+rempli);
affichage(polygone(K,P3,N),1+rempli);
affichage(polygone(B,L,P1,M1),2+rempli);
affichage(polygone(N,P2,J,K),2+rempli);
affichage(polygone(B,L,K),3+rempli);
affichage(polygone(B,J,K),3+rempli);
affichage(polygone(P,L,K),4+rempli);
affichage(polygone(D,P2,J),4+rempli);
affichage(polygone(P,A,K),5+rempli);
affichage(polygone(P3,A,K),5+rempli);
K:=K;
B:=B;
P1:=P1;
P2:=P2;
P3=P3;
N:=N;
L:=L;
J:=J;
segment(A,B);
M1:=M1;
```

11.3 Puzzle transformant un triangle en un triangle de base et de hauteur inégales

11.3.1 Le principe

Soient 2 triangles ABC et DCE ayant la même aire mais de base et de hauteur inégales.
On va transformer le triangle ABC en un triangle A_1BC de même aire et vérifiant $A_1C = CE$. Le point A_1 est donc l'intersection du cercle de centre C et de rayon CE avec la droite $y = -h_1$ si h_1 est la longueur de la hauteur issue de A du triangle ABC.
On va transformer ensuite transformer par rotation de centre C le triangle A_1BC en le triangle D_1EC. Le point D_1 est donc l'intersection du cercle de centre C et de rayon CE avec la droite $y = -h_2$ si h_2 est la longueur de la hauteur issue de D du triangle DCE.



Ainsi on peut trouver un puzzle qui transforme ABC en A_1BC puis A_1BC est transformé par rotation en D_1CE, puis on trouve un puzzle qui transforme D_1CE en DCE et en coupant certaines pièces ont fait en sorte d'avoir les mêmes pièces pour les 2 puzzles.

11.3.2 Les noms des différents points

Les noms dont on va avoir besoin :

```
A:=point(-7+3*i));
B:=point(-8);
C:=point(0);
D:=point(4+4*i);
E:=point(6,affichage=quadrant4);
d:=droite(y=-3,affichage=2+ligne_tiret_point):;d;
d1:=droite(y=-4,affichage=2+ligne_tiret_point):;d1;
droite(y=3,affichage=2+ligne_tiret_point);
droite(y=4,affichage=2+ligne_tiret_point);
A1:=inter_unique(cercle(C,longueur(C,E)),d,A);
D1:=inter_unique(cercle(C,longueur(C,B)),d1,D);
c:=evalf(angle(C,A1,E));
D1:=rotation(C,c,B);
triangle(A,B,C);
triangle(A1,B,C);
triangle(D1,C,E);
triangle(D,C,E);
K:=inter_unique(droite(A,A1),droite(B,C));
K1:=inter_unique(droite(D,D1),droite(B,C));
L:=milieu(A,B);segment(K,L);
```

```
M:=milieu(A,C);segment(K,M);
N:=milieu(C,A1);segment(K,N);
P:=milieu(B,A1);segment(K,P);
Q:=milieu(D,C);segment(K1,Q);
R:=milieu(D,E);segment(K1,R);
S:=milieu(D1,E);segment(K1,S);
T:=milieu(D1,C);segment(K1,T);
N1:=milieu(C,E);
K2:=rotation(C,c,K);;
segment(K2,S);
segment(K2,N1);
K3:=translation(D-K1,K2);
U:=inter_unique(segment(T,K1),segment(N1,K2));
U1:=translation(D-K1,U);
segment(U1,K3);
segment(R,K3);
N2:=symetrie(milieu(C,K1),N1,affichage=quadrant2);
U2:=symetrie(milieu(C,K1),U);
segment(U2,N2);
U3:=rotation(C,-c,U);
T1:=milieu(B,C);
K4:=rotation(C,-c,K1);
segment(K4,P);
segment(K4,T1);
U4:=translation(A-K,U3);
K5:=translation(A-K,K4);
segment(K5,U4);
segment(K5,L);
T2:=symetrie(milieu(K,C),T1);
U5:=symetrie(milieu(K,C),U3);
segment(U5,T2);
```

On obtient :



11.3.3 Le puzzle obtenu

On obtient 6 pièces.
On tape :

```
affichage(polygone(A,L,K5,U4),1+rempli);
affichage(triangle(L,K,K5),2+rempli);
affichage(polygone(K,P,K4,U3),1+rempli);
affichage(polygone(K1,U,K2,S),1+rempli);
affichage(polygone(D,U1,K3,R),1+rempli);
affichage(triangle(P,A1,K4),2+rempli);
affichage(triangle(E,K1,S),2+rempli);
affichage(triangle(E,K1,R),2+rempli);
affichage(triangle(B,L,K),3+rempli);
affichage(triangle(B,P,K),3+rempli);
affichage(triangle(D1,S,K2),3+rempli);
affichage(triangle(R,K3,K1),3+rempli);
affichage(triangle(U4,K5,M),4+rempli);
affichage(triangle(K4,U3,N),4+rempli);
affichage(triangle(K1,U,N1),4+rempli);
affichage(triangle(C,U2,N2),4+rempli);
affichage(polygone(M,K,T2,U5),5+rempli);
affichage(polygone(N,C,T1,U3),5+rempli);
affichage(polygone(C,N1,U,T),5+rempli);
```

```
affichage(polygone(Q,U2,N2,K1),5+rempli);
affichage(triangle(C,U5,T2),6+rempli);
affichage(triangle(K,U3,T1),6+rempli);
affichage(triangle(K2,U,T),6+rempli);
affichage(triangle(Q,U1,K3),6+rempli);
A:=A;A1:=A1;
D:=D;D1:=D1;
B:=B;
C:=C;
E:=E;
```

On obtient le puzzle des 4 triangles :



11.4 Puzzle transformant un triangle équilatéral en un carré

On transforme facilement un carré en un triangle rectangle de même aire avec un puzzle de 2 pièces.



On va donc transformer un carré en un triangle rectangle et avec la méthode ci-dessus, transformer ce triangle rectangle en un triangle équilatéral de même aire. Si le carré est de côté a, les côtés de l'angle droit du triangle rectangle sont a et $2a$ et les côtés du triangle équilatéral de même aire sont $b = 2a * 3^{-1/4}$

```
a:=1;b:=2*a/sqrt(sqrt(3.));
A:=point(i*a);B:=point(-2*a);
C:=point(0);E:=point(b);
D:=point(b/2+i*b*sqrt(3)/2);
triangle_equilateral(0,b);
triangle(0,i*a,-2*a);
A1:=inter_unique(cercle(C,b),droite(y=-a),B);
triangle(A1,B,C);
K:=inter_unique(droite(A,A1),droite(B,C));
L:=milieu(A,B);
M:=milieu(A,C);
N:=milieu(A1,C);
P:=milieu(A1,B);
segment(K,L);
segment(K,P);
segment(K,M);
segment(K,N);
D1:=inter_unique(cercle(C,longueur(C,B)),
               droite(y=-b*sqrt(3)/2),D);
triangle(D1,C,E);;
K1:=inter_unique(droite(D,D1),droite(B,C));
Q:=milieu(D,C);segment(K1,Q);
R:=milieu(D,E);segment(K1,R);
S:=milieu(D1,E);segment(K1,S);
T:=milieu(D1,C);segment(K1,T);
N1:=milieu(C,E);
c:=evalf(angle(C,A1,E));
K2:=rotation(C,c,K);;
segment(K2,S);
segment(K2,N1);
U:=inter_unique(segment(S,K2),segment(T,K1));
U1:=symetrie(milieu(C,K1),U);
K3:=symetrie(milieu(C,K1),K2);
segment(U1,K3);
U2:=translation(D-K1,U);
segment(U2,R);
N2:=symetrie(milieu(C,K1),N1);
segment(N2,K3);
U3:=rotation(C,-c,U);
K4:=rotation(C,-c,K1);
T1:=milieu(B,C);
segment(T1,K4);
segment(P,K4);
U4:=translation(A-K,U3);
```

```
K5:=translation(A-K,K4);
segment(U4,K5);
segment(L,K5);
T2:=symetrie(milieu(B,K),T1);
U5:=symetrie(milieu(B,K),U3);
segment(T2,U5);
T3:=symetrie(L,T2)
T4:=point(-a+i*a)
segment(T1,T4)
segment(U4,T3)
R1:=projection(droite(K1,Q),R)
segment(R,R1)
segment(T4,i)
```




```
affichage(polygone(A,M,K5,U4),1+rempli);
affichage(triangle(L,U4,K5),2+rempli);
affichage(polygone(K,N,K4,U3),1+rempli);
affichage(polygone(K1,U,K2,N1),1+rempli);
affichage(polygone(C,U1,K3,N2),1+rempli);
affichage(triangle(P,U3,K4),2+rempli);
affichage(triangle(U,K1,S),2+rempli);
affichage(triangle(D,U2,R),2+rempli);
affichage(triangle(B,T2,U5),3+rempli);
affichage(triangle(T1,U3,K),3+rempli);
affichage(triangle(U,T,K2),3+rempli);
affichage(triangle(Q,K3,U1),3+rempli);
affichage(triangle(C,K,M),4+rempli);
affichage(triangle(K,C,N),4+rempli);
affichage(triangle(K2,C,N1),4+rempli);
```

```
affichage(triangle(K1,N2,K3),4+rempli);
affichage(polygone(L,K,T2,U5),5+rempli);
affichage(polygone(P,B,T1,U3),5+rempli);
affichage(polygone(D1,S,U,T),5+rempli);
affichage(polygone(R,U2,Q,K1),5+rempli);
affichage(triangle(L,K,K5),6+rempli);
affichage(triangle(K4,P,A1),6+rempli);
affichage(triangle(K1,E,S),6+rempli);
affichage(triangle(E,R,K1),6+rempli);
A:=A;A1:=A1;
D:=D;D1:=D1;
B:=B;
C:=C;
E:=E;
```



On obtient 7 pièces :

```
affichage(polygone(C,U1,K3,N2),1+rempli);
affichage(triangle(D,U2,R),2+rempli);
affichage(triangle(Q,K3,U1),3+rempli);
affichage(triangle(K1,N2,K3),4+rempli);
affichage(triangle(D,U2,R),2+rempli);
affichage(polygone(R,U2,Q,R1),5+rempli);
affichage(polygone(R1,R,K1),47+rempli);
affichage(triangle(E,R,K1),6+rempli);
affichage(polygone(A,M,K5,U4),1+rempli);
affichage(triangle(L,U4,K5),2+rempli);
affichage(triangle(i,T3,U4),3+rempli);
affichage(triangle(C,K,M),4+rempli);
```

```
affichage(polygone(L,K,T1),47+rempli);
affichage(polygone(L,U4,T3,T4),5+rempli);
affichage(triangle(L,K,K5),6+rempli);
```



On voit que l'on peut reunir la pièce rouge avec la jaune la pièce magenta avec la verte et la pièce grise avec la cyan.

```
affichage(polygone(C,U1,K3,N2),1+rempli);
affichage(triangle(D,U2,R),2+rempli);
affichage(triangle(Q,K3,U1),1+rempli);
affichage(triangle(K1,N2,K3),4+rempli);
affichage(polygone(R,U2,Q,R1),2+rempli);
affichage(polygone(R1,R,K1),6+rempli);
affichage(triangle(E,R,K1),6+rempli);
affichage(polygone(A,M,K5,U4),1+rempli);
affichage(triangle(L,U4,K5),2+rempli);
affichage(triangle(i,T3,U4),1+rempli);
affichage(triangle(C,K,M),4+rempli);
affichage(polygone(L,K,T1),6+rempli);
affichage(polygone(L,U4,T3,T4),2+rempli);
affichage(triangle(L,K,K5),6+rempli);
```

On obtient donc 4 pièces :



11.5 Puzzle transformant un rectangle 5x1 en un carré

On tape :

```
rectangle(-5,0,1/5):;
carre(-5,-4,affichage=1+rempli);
triangle(-4,-2,-2+i,affichage=2+rempli);
triangle(-4,-2+i,-4+i,affichage=3+rempli);
triangle(-2,0,-2+i,affichage=4+rempli);
triangle(0,i,-2+i,affichage=5+rempli);
carre(-3-5/2*i,-2-5/2*i, affichage=1+rempli));
triangle(-4-3/2*i,-2-3/2*i,-2-i/2,affichage=2+rempli);
triangle(-4-3/2*i,-3-3/2*i,-3-7/2*i,affichage=3+rempli);
triangle(-1-5/2*i,-3-5/2*i,-3-7/2*i,affichage=4+rempli);
triangle(-1-5/2*i,-2-5/2*i,-2-i/2,affichage=5+rempli);
```

On obtient les 5 pièces du puzzle du rectangle $5 * 1$:



11.6 Puzzle transformant un rectangle 3x2 en un carré

Soit un rectangle de longueur 3 unités et de largeur 2 unités.
On veut le partager en plusieurs morceaux pour pouvoir constituer avec tous ces morceaux un carré.
On prend exemple sur le découpage précédent.
Le carré doit avoir comme côté $\sqrt{6}$ unités.
$\sqrt{6}$ est l'hypoténuse d'un triangle rectangle de côté 2 et $\sqrt{2}$ (puisque $6 = 4 + 2 = 2^2 + (\sqrt{2})^2 = (\sqrt{6})^2$.
On utilise la formule :
$4 * \sqrt{2} + (2 - \sqrt{2})^2 = 6$ et
On considère la droite D d'équation $y = -\sqrt{2} * x + 2$ (c'est la droite portant l'hypoténuse du triangle $0, \sqrt{2}, 2 * i$.
Puisque le point de coordonnées $(1, 2 - \sqrt{2})$ est sur la droite D, on peut découper le rectangle selon les 5 morceaux ci-dessous.
Voici ce puzzle fait avec ces 5 morceaux mais on remarquera qu'il suffit de découper le rectangle en seulement 3 morceaux ! ! !
On tape :

```
rectangle(0,3,2/3):;
T1:=triangle(0,sqrt(2),2i):; affichage(T1,1+rempli);
T2:=triangle(1+2*i,1+(2-sqrt(2))*i,3+2*i):;
affichage(T2,2+rempli);
T3:=triangle(3+2*i,1+(2-sqrt(2))*i,3+(2-sqrt(2))*i):;
affichage(T3,3+rempli);
T4:=triangle(2i,1+(2-sqrt(2))*i,1+2i):;
affichage(T4,4+rempli);
```

```
T5:=polygone(1+(2-sqrt(2))*i,sqrt(2),1+sqrt(2),
             1+sqrt(2)+(2-sqrt(2))*i):;
affichage(T5,5+rempli);
T6:=carre(1+sqrt(2),3):;
affichage(T6,6+rempli);
carre(sqrt(2)-2,2i-2):;
affichage(translation(-2,T1),1+rempli);
affichage(translation(-5,T3),3+rempli);
affichage(translation(-5,T5),5+rempli);
affichage(translation(-5,T6),6+rempli);
affichage(translation(-3+sqrt(2)-2-2i,T2),2+rempli);
affichage(translation(-3+sqrt(2)-2-2i,T4),4+rempli);
```

On obtient les 5 pièces du puzzle du rectangle $3 * 2$:



On obtient avec 3 pièces :



Autre solution Soit un rectangle de longueur 3 unités et de largeur 2 unités.
On veut le partager en plusieurs morceaux pour pouvoir constituer avec tous ces morceaux un carré.
Il reste ensuite à découper le rectangle $(3-\sqrt{5})*2$ en 3 morceaux pour en faire un carré selon la méthode de la section 11.9.
rappelons ce découpage sur cet exemple.
On tape :

```
b:=3;a:=2;
A:=point(0);
B:=point(b);
C:=point(b+i*a);
D:=point(i*a);
rectangle(A,B,C,D);
P:=point(sqrt(a*b));
M:=point(b-sqrt(a*b)+i*a);
Q:=point(sqrt(a*b)*(1+i));
R:=point(sqrt(a*b)*i);
S:=point(sqrt(a*b)+i*a);
T:=point(sqrt(a*b)+i*(sqrt(a*b)-a));
carre(A,P,Q,R, affichage=rouge);
segment(P,D);
segment(B,M);
segment(R,M,affichage=bleu);
segment(P,S, affichage=vert);
```

On obtient :



Les triangles RQP et MCB sont égaux,
les triangles RDM et TPB sont égaux, donc les 3 pièces du puzzle sont les triangles MCB et TPB et le polygone $APTMD$.

Puis on tape :

```
c:=3;
P1:=polygone(A,P,T,M,D,affichage=1+rempli);
T2:=triangle(T,P,B,affichage=2+rempli);
T4:=triangle(M,C,B,affichage=4+rempli);
affichage(translation(-c*i,P1),1+rempli)
affichage(translation(-sqrt(a*b)-c*i+a*i,T2),2+rempli);
affichage(translation(sqrt(a*b)-b-c*i+(sqrt(a*b)-a)*i,T4),
          4+rempli);
```

On obtient les 3 pièces du puzzle du rectangle $3 * 2$:



Autre solution

Soit un rectangle de longueur 3 unités et de largeur 2 unités.
On veut le partager en plusieurs morceaux pour pouvoir constituer avec tous ces morceaux un carré.
On prend exemple sur le découpage précédent.
Le carré doit avoir comme côté $\sqrt{6}$ unités.
$\sqrt{6}$ est l'hypoténuse d'un triangle rectangle de côté 1 et $\sqrt{5}$ (puisque $6 = 1 + 5 = 1^2 + (\sqrt{5})^2 = (\sqrt{6})^2$.
On utilise la formule :
$2 * \sqrt{5} + (\sqrt{5} - 1)^2 = 6$ et
Dans le rectangle de côtés 1 et $2 * \sqrt{5}$, on découpe 4 triangles rectangles qui ont des côtés de l'angle droit de longueur et 1 et $\sqrt{5}$.
Il reste ensuite à découper le rectangle $(3 - \sqrt{5}) * 2$ en 3 morceaux pour en faire un carré selon la méthode de la section 11.9.
Rappelons ce découpage sur cet exemple.
On tape :

```
b:=2;a:=3-sqrt(5);
A:=point(0);
B:=point(b);
C:=point(b+i*a);
D:=point(i*a);
rectangle(A,B,C,D);
P:=point(sqrt(a*b));
M:=point(b-sqrt(a*b)+i*a);
Q:=point(sqrt(a*b)*(1+i));
R:=point(sqrt(a*b)*i);
S:=point(sqrt(a*b)+i*a);
T:=point(sqrt(a*b)+i*(sqrt(a*b)-a));
carre(A,P,Q,R, affichage=rouge);
segment(P,D);
segment(B,M);
segment(R,M,affichage=bleu);
segment(P,S, affichage=vert);
```

Puis on tape :

```
P1:=rotation(0,pi/2,polygone(A,P,T,M,D)):;
affichage(P1,1+rempli);
T2:=rotation(0,pi/2,triangle(T,P,B)):;
affichage(T2,2+rempli);
T4:=rotation(0,pi/2,triangle(M,C,B)):;
affichage(T4,4+rempli);
affichage(translation(2,P1),1+rempli);
affichage(translation(-1+sqrt(5)-sqrt(5)*i+i,T2),2+rempli);
affichage(translation(6-2*sqrt(5)+(sqrt(5)-3)*i,T4),4+rempli);
```

On obtient avec les 3 pièces qui formeront le carré central :



Pour faire les 7 morceaux, on tape :

```
rectangle(0,3,2/3):;
T1:=triangle(0,sqrt(5),i):;
affichage(T1,1+rempli);
T2:=triangle(i+sqrt(5),sqrt(5),i):;
affichage(T2,2+rempli);
T3:=triangle(i,sqrt(5)+i,2*i):;
affichage(T3,3+rempli);
T4:=triangle(2*i,sqrt(5)+2*i,i+sqrt(5)):;
affichage(T4,4+rempli);
T5:=polygone(sqrt(5),3, 3+i*(sqrt(5)-1),
    i*(sqrt(5)-1)+7-2*sqrt(5), i*(3-sqrt(5))+sqrt(5)):;
affichage(T5,5+rempli);
T6:=triangle(3+2*i,2*i+sqrt(5),sqrt(5)+i*(3-sqrt(5))):;
affichage(T6,6+rempli);
T7:=triangle(3+i*(sqrt(5)-1),3+2*i,i*(sqrt(5)-1)+7-2*sqrt(5)):;
affichage(T7,47+rempli);
affichage(translation(-2*i-i/2,T5),5+rempli);
affichage(translation(4-2*sqrt(5)+sqrt(5)*i-5*i-i/2,T6),6+rempli);
affichage(translation(-3+sqrt(5)-sqrt(5)*i-i-i/2,T7),47+rempli);
affichage(translation(sqrt(5)*i-3*i-sqrt(5)+4-i/2,T1),1+rempli);
affichage(translation(-sqrt(5)+3-3*i-i/2,T2),2+rempli);
affichage(translation(-3*i-sqrt(5)+4-i/2,rotation(i,pi/2,T3)),
          3+rempli));
affichage(translation(3-sqrt(5)+sqrt(5)*i-5*i-i/2,
                rotation(sqrt(5)+2*i,pi/2,T4)),4+rempli));
```

On obtient les 7 pièces du puzzle du rectangle $3 * 2$:



Autre solution
On fait 2 triangles rectangles ($T1etT2$) de côtés $\sqrt{2}$,2,$\sqrt{6}$, puis on fait 2 triangles rectangle ($T5etT6$) pour constituer le carré central de côté $2 - \sqrt{2}$. Puis on complète en respectant la symétrie. On obtient 10 morceaux, on tape :

```
rectangle(0,3,2/3):;
T1:=triangle(0,2,sqrt(2)*i):;
affichage(T1,1+rempli);
T2:=triangle(3+2*i,1+2*i,3+(2-sqrt(2))*i):;
affichage(T2,2+rempli);
T3:=triangle(sqrt(2)*i+2-sqrt(2),sqrt(2)*i,2):;
affichage(T3,3+rempli);
T4:=triangle(1+2*i,1+sqrt(2)+(2-sqrt(2))*i,3+(2-sqrt(2))*i):;
affichage(T4,4+rempli);
T5:=triangle(sqrt(2)*i,2*i,2-sqrt(2)+sqrt(2)*i):;
affichage(T5,5+rempli);
T6:=triangle(3+(2-sqrt(2))*i,3,(2-sqrt(2))*i+sqrt(2)+1):;
affichage(T6,6+rempli);
T7:=triangle(2*i,1+2*i,1+i):;
affichage(T7,47+rempli);
T8:=triangle(1+2*i,1+i,2+i):;
affichage(T8,178+rempli);
T9:=triangle(2+i,1+i,2):;
affichage(T9,69+rempli);
T0:=triangle(2+i,3,2):;
affichage(T0,80+rempli);
affichage(translation(4.5,T1),1+rempli);
affichage(translation(4.5,
```

```
          rotation(sqrt(2)*i,-pi/2,T3)),3+rempli);
affichage(translation(4.5-2*i,T5),5+rempli);
affichage(translation(3.5-sqrt(2)-(2-sqrt(2))*i,T6),6+rempli);
affichage(translation(3.5-sqrt(2)-4*i+sqrt(2)*i,T2),2+rempli);
affichage(translation(3.5-sqrt(2)-4*i+sqrt(2)*i,
          rotation(3+(2-sqrt(2))*i,-pi/2,T4)),4+rempli);
affichage(translation(3.5-sqrt(2)/2-4*i+3*sqrt(2)/2*i,
          rotation(1+2*i,pi/4,T7)),47+rempli);
affichage(translation(3.5-sqrt(2)/2-3*i+3*sqrt(2)/2*i,
          rotation(1+i,-pi/4,T8)),178+rempli);
affichage(translation(4.5-sqrt(2)/2-i-sqrt(2)/2*i,
          rotation(2+i,-pi/4,T9)),69+rempli);
affichage(translation(4.5-sqrt(2)/2-sqrt(2)/2*i,
          rotation(2,pi/4,T0)),80+rempli);
```

On obtient les 10 pièces du puzzle du rectangle $3 * 2$:



11.7 Puzzle transformant un rectangle $a * (a + 1)$ en un carré

Soit un rectangle de longueur $a + 1$ unités et de largeur a unités avec $a \geq 1$ (pour avoir $\sqrt{a} \leq a$).
On veut le partager en plusieurs morceaux pour pouvoir constituer avec tous ces morceaux un carré.
On prend exemple sur le découpage précédent.
Le carré reconstitué a pour côté $\sqrt{a^2 + a}$ unités.
$\sqrt{a^2 + a}$ est l'hypoténuse d'un triangle rectangle de côté a et $\sqrt{a}$ (puisque $a^2+a = a^2 + (\sqrt{a})^2 = (\sqrt{a^2 + a})^2$.
On utilise la formule :

$2a*\sqrt{a}+(a-\sqrt{a})^2=a^2+a$
On considère la droite D d'équation $y=-\sqrt{a}x+a$ (c'est la droite portant l'hypoténuse du triangle $0,\sqrt{a},a*i$
. Puisque le point de coordonnées $(1,a-\sqrt{a})$ est sur la droite D, on peut découper le rectangle selon les 5 morceaux ci-dessous.
Voici ce puzzle fait avec ces 5 morceaux mais on remarquera qu'il suffit de découper le rectangle en seulement 3 morceaux ! ! !
On tape :

```
supposons(a=[2,1.0,10.0,0.1]);
rectangle(0,a+1,a/(a+1)):;
T1:=triangle(0,sqrt(a),a*i):; affichage(T1,1+rempli);
T2:=triangle(1+a*i,1+(a-sqrt(a))*i,a+1+a*i):;
affichage(T2,2+rempli);
T3:=triangle(a+1+a*i,1+(a-sqrt(a))*i,a+1+(a-sqrt(a))*i):;
affichage(T3,3+rempli);
T4:=triangle(a*i,1+(a-sqrt(a))*i,1+a*i):;
affichage(T4,4+rempli);
T5:=polygone(1+(a-sqrt(a))*i,sqrt(a),1+sqrt(a),
    1+sqrt(a)+(a-sqrt(a))*i):;
affichage(T5,5+rempli);
T6:=carre(1+sqrt(a),a+1):;
affichage(T6,6+rempli);
carre(sqrt(a)-a,a*i-a):;
affichage(translation(-a,T1),1+rempli);
affichage(translation(-2a-1,T3),3+rempli);
affichage(translation(-2a-1,T5),5+rempli);
affichage(translation(-2a-1,T6),6+rempli);
affichage(translation(-a-1+sqrt(a)-a-a*i,T2),2+rempli);
affichage(translation(-1-a+sqrt(a)-a-a*i,T4),4+rempli);
```

On obtient les 5 (ou 3) pièces du puzzle du rectangle $(a+1)*a$ pour $a=4$:





On obtient les 5 (ou 3) pièces du puzzle du rectangle $(a+1)*a$ pour $a=10$:



11.8 Puzzle transformant un rectangle 3x1 en un carré

Soit un rectangle de longueur 3 unités et de largeur 1 unité.
On veut le partager en plusieurs morceaux pour pouvoir constituer avec tous ces morceaux un carré.
Le carré reconstitué a pour côté $\sqrt{3}$ unités.

11.8.1 On prend exemple sur le découpage précédent

On utilise la formule :
$2 * \sqrt{2} + (\sqrt{2} - 1)^2 = 6$
Donc le carré sera au centre a un côté de longueur $\sqrt{2} - 1$ unités.
Voici ce puzzle fait avec ces 8 morceaux mais on remarquera qu'il suffit de découper le rectangle en seulement 4 morceaux ! ! !
On tape :

```
rectangle(0,3,1/3):;
T1:=triangle(0,sqrt(2),i):;
affichage(T1,1+rempli);
T2:=triangle(3,3-sqrt(2)/2,3+i):;
affichage(T2,2+rempli);
T3:=polygone(2,2+i*(2-sqrt(2)),(2-sqrt(2))*(1+i),sqrt(2)):;
affichage(T3,3+rempli);
T4:=polygone(2,3-sqrt(2)/2,3+i,2+i):;
affichage(T4,4+rempli);
T5:=carre(2+i*(2-sqrt(2)),2+i):;
affichage(T5,5+rempli);
T6:=triangle(i,(2-sqrt(2))*(1+i),2-sqrt(2)+i):;
affichage(T6,6+rempli);
T0:=triangle(i+(2-sqrt(2))*3/2,(2-sqrt(2))*(1+i),2-sqrt(2)+i):;
```

```
affichage(T0,rempli);
T7:=polygone((i+(2-sqrt(2))*3/2,(2-sqrt(2))*(1+i),
               3-sqrt(2)+(2-sqrt(2))*i),3-sqrt(2)+i):;
affichage(T7,47+rempli);
affichage(translation(2-2*i+i/2,T1),1+rempli);
affichage(translation(-1-2*i+i/2,T2),2+rempli);
affichage(translation(-3i-1+i/2+sqrt(2),T7),47+rempli);
affichage(translation(-3i-1+sqrt(2)+i/2,T5),5+rempli);
affichage(translation(-1+sqrt(2)-3*i+i/2,T4),4+rempli);
affichage(translation(-1+sqrt(2)-3*i+i/2,T3),3+rempli);
affichage(translation(-1+2*sqrt(2)-4*i+i/2,T6),6+rempli);
affichage(translation(-1+2*sqrt(2)-4*i+i/2,T0),0+rempli);
```

On obtient les 8 pièces du puzzle du rectangle $3 * 1$:



On obtient avec 5 pièces :



11.8.2 Autre découpage

On construit donc un triangle rectangle `T1` ayant comme côté 1 et $\sqrt{2}$: son hypothénuse a pour longueur $\sqrt{3}$.
`T1` est le triangle PMR.
Puis on trace la perpendiculaire NM à PQ. `T2` est le triangle NMR.
Le triangle `T3` (resp `T4`) est le symétrique de `T1` (resp `T2`) par rapport au centre du rectangle.
Le triangle OPR et son symétrique par rapport au centre du rectangle sont des triangles rectangles isocèles (leurs côtés sont de longueur 1,1 et $\sqrt{2}$).
Le triangle NMQ et son symétrique par rapport au centre du rectangle sont des triangles rectangles isocèles (leurs côtés sont de longueur $\sqrt{2}-1$, $\sqrt{2}-1$, 2-$\sqrt{2}$).
On tape :

```
rectangle(-4,-1,1/3);
segment(-2,-2+i,affichage=ligne_tiret_pointpoint);
segment(-3,-3+i,affichage=ligne_tiret_pointpoint);
T1:=triangle(-4+i,-3,-3+(1+i)/sqrt(2));
O:=point(-4);
M:=point(-3+(1+i)/sqrt(2),affichage=quadrant4);
N:=point(-4+sqrt(2)+i,affichage=quadrant2);
P:=point(-3);
Q:=point(-2+i,affichage=quadrant2);
R:=point(-4+i,affichage=quadrant2);
T2:=triangle(-4+i,-3+(1+i)/sqrt(2),-4+sqrt(2)+i);
T3:=symetrie(point(-5/2+i/2),T1);
T4:=symetrie(point(-5/2+i/2),T2);
```

On obtient les 8 pièces triangulaires qui pavent le rectangle $3 * 1$:



On réalise le puzzle.
On tape :

```
rectangle(-4,-1,1/3):;
T1:=triangle(-4+i,-3,-3+(1+i)/sqrt(2)):;T1;
T2:=triangle(-4+i,-3+(1+i)/sqrt(2),-4+sqrt(2)+i):;
T3:=symetrie(point(-5/2+i/2),T1):;T3;
T4:=symetrie(point(-5/2+i/2),T2):;
translation(3/2-3/2*i,T1,affichage=1+rempli);
R2:=rotation(-3+(1+i)/sqrt(2),pi/2,T2):;T2;
translation(3/2-3/2*i,R2,affichage=2+rempli);
A:=translation(3/2-3/2*i,rotation(-3+(1+i)/sqrt(2),pi/2,-4+i)):;
translation(affixe(A)+1,T3,affichage=3+rempli);
R4:=rotation(-2-1/sqrt(2)+i*(1-1/sqrt(2)),pi/2,T4):;
translation(affixe(A)+1,R4,affichage=4+rempli);
triangle(A,A+sqrt(2)*i,A+(i-1)/sqrt(2),affichage=5+rempli);
B:=translation(affixe(A)+1,rotation(-2-1/sqrt(2)+i*(1-1/sqrt(2)),
             pi/2,point(-1))):;
triangle(B,B-i*sqrt(2),B+(1-i)/sqrt(2),affichage=6+rempli);
segment(B+(1-i)/sqrt(2),A+(i-1)/sqrt(2));
affichage(T1,1+rempli);
affichage(T2,2+rempli);
affichage(T3,3+rempli);
affichage(T4,4+rempli);
T5:=triangle(-1,-1+i,-2+i):;
affichage(T5,5+rempli);
T6:=triangle(-4,-3,-4+i):;
affichage(T6,6+rempli);
T7:=triangle(-3,-1-sqrt(2),-2-1/sqrt(2)+i*(1-1/sqrt(2))):;
affichage(T7,47+rempli);
T0:=triangle(-2+i,-4+sqrt(2)+i,-3+1/sqrt(2)+i*(1/sqrt(2))):;
affichage(T0,rempli);
```

```
triangle(-3/2-3*i/2,B+(1-i)/sqrt(2),A+(i-1)/sqrt(2),affichage=rempli);
triangle(-7/2+sqrt(2)-3*i/2,B+(1-i)/sqrt(2),A+(i-1)/sqrt(2),
          affichage=47+rempli);
```

On obtient les 8 pièces du puzzle du rectangle $3 * 1$:



11.8.3 Autre découpage

On utilise la formule :
$\sqrt(3)^2 + 1^2 = 2^2$
Soit $ABCD$ un rectangle de dimension $1 * 3$.
Soit $APQR$ un carré de dimension $\sqrt{3}$.
On dessine donc 2 triangles rectangles APD et BCM dont les côtés de l'angle droit ont pour longueur 1 et $\sqrt{3}$.
On tape :

```
A:=point(0);
B:=point(3);
C:=point(3+i);
D:=point(i);
rectangle(A,B,C,D);
P:=point(sqrt(3));
M:=point(3-sqrt(3)+i);
Q:=point(sqrt(3)*(1+i));
R:=point(sqrt(3)*i);
S:=point(sqrt(3)+i);
```

```
T:=point(sqrt(3)+i*(sqrt(3)-1));
carre(A,P,Q,R, affichage=rouge);
segment(P,D);
segment(B,M);
segment(R,M,affichage=bleu);
segment(P,S, affichage=vert);
```

On obtient :



Les points B, M, R sont alignés car les droites BR et BM ont même pente qui vaut $\frac{\sqrt{3}}{3} = \frac{1}{\sqrt{3}}$.
Les triangles rectangles BCM et TQR sont égaux ($CM = QR = \sqrt{3}$ et leurs angles sont égaux).
Le point T intersection de BR et PQ a pour coordonnées :
$\sqrt{3}, \sqrt{3} - 1$ puisque :
$PT/PB = \frac{PT}{3-\sqrt{3}} = AR/AB = \frac{\sqrt{3}}{3}$ on a $PT = \sqrt{3} - 1$
Les triangles rectangles PBT et DMR sont donc égaux ($PT = DR = \sqrt{3} - 1$ et leurs angles sont égaux).
On tape :

```
P1:=polygone(A,P,T,M,D,affichage=1+rempli);
T2:=triangle(T,P,B,affichage=2+rempli);
T4:=triangle(M,C,B,affichage=4+rempli)
affichage(translation(-2*i,P1),1+rempli)
affichage(translation(-sqrt(3)-i,T2),2+rempli);
affichage(translation((sqrt(3)-3)*(1+i),T4),4+rempli);
```

On obtient donc 3 pièces :



On peut faire 4 pièces.
On tape :

```
affichage(triangle(A,P,D),3+rempli);
affichage(translation(-2*i,triangle(A,P,D)),3+rempli);
```

On obtient les 3 pièces du puzzle du rectangle $3 * 1$:



Dans la section suivante on fait une généralisation de ce découpage à condition que la longueur du rectangle soit inférieure à 4 fois sa largeur.

11.9 Puzzle transformant un rectangle $a * b$ en un carré

Soit un rectangle de longueur b unités et de largeur a unités avec $a \leq b \leq 4a$.
On veut le partager en plusieurs morceaux pour pouvoir constituer avec tous ces morceaux un carré.
On prend exemple sur le découpage précédent du rectangle 1x3.
Le carré reconstitué a pour côté $\sqrt{ab}$ unités.
On considère le rectangle $ABCD$ ($AD = a$ et $AB = b$.
Soit $APQR$ un carré de dimension $\sqrt{ab}$.
On dessine donc 2 triangles rectangles APD et BCM ayant comme côtés de l'angle droit a et $\sqrt{ab}$.
On tape :

```
supposons(a=[2,0.0,10.0,0.1]);
supposons(b=[2,a,4*a.0,0.1]);
A:=point(0);
B:=point(b);
C:=point(b+i*a);
D:=point(i*a);
rectangle(A,B,C,D);
P:=point(sqrt(a*b));
M:=point(b-sqrt(a*b)+i*a);
Q:=point(sqrt(a*b)*(1+i));
R:=point(sqrt(a*b)*i);
S:=point(sqrt(a*b)+i*a);
T:=point(sqrt(a*b)+i*(sqrt(a*b)-a));
carre(A,P,Q,R, affichage=rouge);
segment(P,D);
segment(B,M);
segment(R,M,affichage=bleu);
segment(P,S, affichage=vert);
```

On obtient les 3 pièces du puzzle du rectangle $b * a$ (pour $a = 1.5$ et $b = 4$) :



Les points BMR sont alignés car les droites BR et BM ont même pente car $\frac{a}{\sqrt{a*b}} = \frac{\sqrt{a*b}}{b}$.
Les triangles rectangles BCM et TQR sont égaux ($CM = QR = \sqrt{ab}$ et leurs angles sont égaux).
Le point T intersection de BR et PQ a pour coordonnées :
$\sqrt{ab}, \sqrt{ab} - a$ puisque :
$PT/PB = \frac{PT}{b-\sqrt{ab}} = AR/AB = \frac{\sqrt{ab}}{b}$ on a $PT = \sqrt{ab} - a$
Pour que T soit sur le segment PS on suppose que $PT = \sqrt{ab} - a \leq a$ c'est à dire $a \leq b \leq 4a$.
Les triangles rectangles PBT et DMR sont donc égaux ($PT = DR = \sqrt{ab} - a$ et leurs angles sont égaux).
On tape en notant c le paramètre qui produit la translation $-c * i$ de la pièce rouge :

```
supposons(a=[2,0.0,10.0,0.1]);
supposons(b=[2,a,4*a.0,0.1]);
supposons(c=[2,0.0,10.0,0.1]);
P1:=polygone(A,P,T,M,D,affichage=1+rempli);
T2:=triangle(T,P,B,affichage=2+rempli);
T4:=triangle(M,C,B,affichage=4+rempli);
affichage(translation(-c*i,P1),1+rempli)
affichage(translation(-sqrt(a*b)-c*i+a*i,T2),2+rempli);
affichage(translation(sqrt(a*b)-b-c*i+(sqrt(a*b)-a)*i,T4),
          4+rempli);
```

On obtient donc 3 pièces du rectangle $b * a$ pour $a = 1.5, b = 3.1, c = 2.5$:



11.10 Exercice

Faire un puzzle qui transforme un rectangle 1*6 en un carré.
Une solution :
Comme 6>4*1, on ne peut pas utiliser le découpage ci-dessus en 3 morceaux. On découpe le rectangle en 2 morceaux de 1*3, puis on fait le découpage en 3 pièces d'un rectangle 2*3 : cela fait 5 pièces pour le rectangle 1*6.
On tape :

```
rectangle(0,3,2/3);
triangle(3,3+2*i,3-sqrt(6)+2*i);
carre(0,sqrt(6));
segment(i*sqrt(6),3);
T1:=polygone(i,0,sqrt(6),sqrt(6)+i*(sqrt(6)-2),i+3-sqrt(6)/2):;
affichage(T1,1+rempli);
T2:=triangle(3,3+i,3-sqrt(6)/2+i):;
affichage(T2,2+rempli);
T3:=triangle(sqrt(6),3,sqrt(6)+i*(sqrt(6)-2)):;
affichage(T3,3+rempli);
T4:=polygone(6,6+i,6+i-sqrt(6),6-sqrt(6)/2):;
affichage(T4,4+rempli);
T5:=polygone(i+3,3,6-sqrt(6)/2,i+6-sqrt(6)):;
affichage(T5,5+rempli);
affichage(translation(-3*i,T1),1+rempli);
affichage(translation(sqrt(6)-3+sqrt(6)*i-5*i,T2),2+rempli);
affichage(translation(-sqrt(6)-i,T3),3+rempli);
affichage(translation(-6+sqrt(6)+sqrt(6)*i-4*i,T4),4+rempli);
affichage(translation(-3-2*i,T5),5+rempli);
```

On obtient donc 5 pièces :



Une autre solution :
On découpe le rectangle en 2 morceaux de 1*3, puis on fait le découpage en 3 pièces d'un rectangle $a*(a+1)$ avec $a=2$: cela fait 6 pièces pour le rectangle 1*6.
On tape :

```
T1:=polygone(0,sqrt(2),sqrt(2)/2+i,i):;
affichage(T1,1+rempli);
T4:=triangle(3,3+sqrt(2)/2,3+i):;
affichage(T4,4+rempli);
T2:=triangle(1+(2-sqrt(2))*i,3-sqrt(2)+i,sqrt(2)/2+i):;
affichage(T2,2+rempli);
T3:=polygone(sqrt(2),3,3+i,3-sqrt(2)+i,1+(2-sqrt(2))*i):;
affichage(T3,3+rempli);
T5:=triangle(6-sqrt(2),6,6+i):;
affichage(T5,5+rempli);
T6:=polygone(6-sqrt(2),6+i,3+i,3+sqrt(2)/2):;
affichage(T6,6+rempli);
affichage(translation(3-2*i-i/2,T1),1+rempli);
affichage(translation(-i-i/2,T4),4+rempli);
affichage(translation(-3+sqrt(2)-3*i-i/2,T6),6+rempli);
affichage(translation(sqrt(2)+-4*i-i/2,T2),2+rempli);
affichage(translation(-2*i-i/2,T3),3+rempli);
affichage(translation(-3-i-i/2,T5),5+rempli);
```

On obtient donc 6 pièces :



Encore une autre solution :
On utilise le découpage en 3 morceaux en le modifant : on rajoute un quatrième morceau pour que le porcédé marche encore quand $4 = 4a < b = 6 < 8a = 8$.
On tape :

```
rectangle(0,6,1/6);
carre(0,sqrt(6));
segment(6,sqrt(6)*i);
segment(i*2,6-2*sqrt(6)+i*2);
segment(2*sqrt(6),2*sqrt(6)+i*(sqrt(6)-2));
T1:=polygone(0,sqrt(6),sqrt(6)+i,i):;
T2:=polygone(sqrt(6)+i,sqrt(6),2*sqrt(6),
                2*sqrt(6)+i*(sqrt(6)-2),6-sqrt(6)+i):;
T3:=triangle(6,6+i,6-sqrt(6)+i):;
T4:=triangle(2*sqrt(6),6,2*sqrt(6)+i*(sqrt(6)-2)):;
affichage(T1,1+rempli);
affichage(T2,2+rempli);
affichage(T3,3+rempli);
affichage(T4,4+rempli);
```

On obtient :



La droite d'équation $y = -\sqrt{6}/6 * x + \sqrt{6}$ détermine un petit triangle rectangle (en noir) qui se trouve à l'extérieur du rectangle et du carré. Ce triangle a pour côté de l'angle droit : $6 - 2\sqrt{6}$ et $\sqrt{6} - 2$. On le translate dans le rectangle (en bleu) et dans le carré. On obtient donc les 4 pièces ci dessus.
On réalise le puzzle avec ces 4 pièces, on tape :

```
affichage(T1,1+rempli);
affichage(T2,2+rempli);
affichage(T3,3+rempli);
affichage(T4,4+rempli);
affichage(translation(1.5*i,T1),1+rempli);
affichage(translation(2.5*i-sqrt(6),T2),2+rempli);
affichage(translation(sqrt(6)*i+0.5*i-6+sqrt(6),T3),3+rempli);
affichage(translation(3.5*i-2*sqrt(6),T4),4+rempli);
```

On obtient :



11.11 Puzzle transformant une croix de St André en un carrè

On tape :

```
C:=polygone(i,0,1,-i+1,-i+2,2,3,3+i ,2+i,2+2*i,1+2*i,1+i);
```

On obtient :



Comment transformer cette croix en un carré ?
Le polygone `C` a une aire de 5 carrés de côté 1 unité.
Donc `C` se transforme en un carré de côté $\sqrt{5}$ unités. $\sqrt{5}$ est l'hypoténuse d'un triangle rectangle de côtés 1 unité et 2 unités.
On tape :

```
C;
carre(0,1-2i);
segment(1-i,1-2*i);
segment(2+i,3/2+2*i);
segment(1/2-i,1-i);
```

On obtient :



Le puzzle
On tape :

```
C1:=polygone(0,1,-i+1,-i+2,2,5/2,2+i):;
affichage(C1,1+rempli);
C2:=polygone(2+i,3/2+2*i,1+2*i,1+i,i,0):;
affichage(C2,2+rempli);
C3:=triangle(2+i,2+2*i,3/2+2*i):;
affichage(C3,3+rempli);
C4:=polygone(2+i,5/2,3,3+i):;
affichage(C4,4+rempli);
affichage(translation(7/2+i,C1),1+rempli);
affichage(translation(7/2+i+1-2*i,C2),2+rempli);
affichage(translation(7/2+i-1-3*i,C3),3+rempli);
affichage(translation(7/2+i-2-i,C4),4+rempli);
```

On obtient :



11.12 Puzzle transformant une autre croix en un carrè

On tape :

```
P:=polygone(i,0,1,-i+1,-i+2,-2*i+2,-2*i+3,-i+3,-i+4,
4,5,5+i,4+i,4+2*i,3+2*i,3+3*i,2+3*i,2+2*i,1+2*i,1+i);
```

On obtient :



Comment transformer cette croix en un carré ?
Le polygone P a une aire de 13 carrés de côté 1 unité.
Donc P se transforme en un carré de côté $\sqrt{13}$ unités. $\sqrt{13}$ est l'hypoténuse d'un triangle rectangle de côtés 2 unités et 3 unités.
On tape :

```
P;
carre(-i,3-3*i);
segment(2+2*i,2+2/3+3*i);
segment(3-2*i,3-3*i);
segment(3-2*i,3+2/3-2*i);
```

On obtient :



Le puzzle
On tape :

```
P1:=polygone(2/3,1,,-i+1,-i+2,-2*i+2,
-2*i+3,-i+3,-i+4,4,5,2+2*i):;
affichage(P1,1+rempli);
P2:=polygone(5,5+i,4+i,
4+2*i,3+2*i,3+3*i,
2+2/3+3*i,2+2*i):;
affichage(P2,2+rempli);
P3:=triangle(2+3*i,2+2/3+3*i,2+2*i):;
```

```
affichage(P3,3+rempli);
P4:=polygone(2+2*i,1+2*i,1+i,i,0,2/3):;
affichage(P4,4+rempli):
affichage(translation(6+i,P1),1+rempli);
affichage(translation(6+i-2-3*i,P2),2+rempli);
affichage(translation(6+i+1-5*i,P3),3+rempli);
affichage(translation(6+i+3-2*i,P4),4+rempli);
```

On obtient :



11.13 Puzzle transformant encore une autre croix en un carrè

On tape :

```
polygone(i,0,1,-i+1,-2*i+1,-2*i+2,
-2*i+2,2,3,3+i,2+i,2+2*i,1+2*i,1+i);
```

On obtient :



Comment transformer cette croix en un carré ?

11.13.1 Une première façon

Cette croix a une aire de 6 carrés de côté 1 unité.
Donc elle se transforme en un carré de côté $\sqrt{6}$ unités. Pour cela on passe par l'intermédiaire d'un rectangle de côtés 1 unité et 6 unités.
On tape :

```
polygone(i,0,1,-i+1,-2*i+1,-2*i+2,
          -2*i+2,2,3,3+i,2+i,2+2*i,1+2*i,1+i);
rectangle(5-3*i,6-3*i,6);
carre(5+3*i-sqrt(6)*i,5+sqrt(6)+3*i-sqrt(6)*i);
segment(5-3*i,5+sqrt(6)+3*i);
segment(7+3*i,7+2*i);
segment(5+i,6+i);
segment(5+2*i,6+2*i);
segment(1,1+i);
segment(2,2+i);
A:=point(i*(3-sqrt(6))+sqrt(6)+4);
B:=point(7+i*(2*sqrt(6)-3));
simplify(B-A),simplify(A-E);
segment(i*(3-2*sqrt(6))+ sqrt(6)+3,5+i*(3-2*sqrt(6)));
segment(4*i-sqrt(6)*i+1,4*i-sqrt(6)*i+2);
D:=point(2+sqrt(6)*i-2*i);
E:=point(6+i*(sqrt(6)-3));
segment(1-2*i,2+(sqrt(6)-2)*i);
segment(i*(4-2*sqrt(6))+ sqrt(6)-1,1+i*(4-2*sqrt(6)));
C:=point(i*(4-2*sqrt(6))+sqrt(6))-1;
F:=point(2+i*(4-sqrt(6)));
```

On obtient 6 pièces :



Le puzzle
On tape :

```
L1:=carre(0,1):;
affichage(L1,1+rempli);
L2:=polygone(1-2*i,2-2*i,D):;
affichage(L2,2+rempli);
L3:=polygone(1+i*(4-sqrt(6.)),2+i*(4-sqrt(6.)),2+2*i,1+2*i):;
affichage(L3,3+rempli);
L4:=polygone(C,D,F,1+i*(4-sqrt(6.)),i*(4-2*sqrt(6.))+1):;
affichage(L4,4+rempli);
L5:=triangle(1-2*i,C,1+i*(4-2*sqrt(6.))):;
affichage(L5,5+rempli);
L6:=carre(2,3):;
affichage(L6,6+rempli);
affichage(translation(4-i,L1),1+rempli);
affichage(translation(2,L6),6+rempli);
affichage(translation(A-point(2),L2),2+rempli);
affichage(translation(3-3*i,L3),3+rempli);
affichage(translation((A-C)-1-2*i,L4),4+rempli);
affichage(translation(5+i*(2*sqrt(6)-3),L5),5+rempli);
```

On obtient :



11.13.2 Une deuxième façon

Cette croix a une aire de 6 carrés de côté 1 unité.
Donc elle se transforme en un carré de côté $\sqrt{6}$ unités. Pour cela on passe par l'intermédiaire d'un rectangle de côtés 2 unités et 3 unités.
On tape :

```
Q:=polygone(i,0,1,-i+1,-2*i+1,-2*i+2, -2*i+2,2,3,3+i,2+i,2+2*i,
            1+2*i,1+i):;
segment(2-2/sqrt(6)-2*i,2-i):;
segment(1+(2-sqrt(6))*i,3-2/sqrt(6)+i):;
segment(1-i,2-i):;
segment(1,1+i):;
Q1:=polygone(1+(2-sqrt(6.))*i,3-2/sqrt(6.)+i,2+i, 2+2*i,1+2*i):;
affichage(Q1,1+rempli);
Q2:=polygone(3-2/sqrt(6.)+i,1+(2-sqrt(6.))*i, 1-i,2-i,2,3,3+i):;
affichage(Q2,2+rempli);
Q3:=polygone(1-i,1-2*i,2-2/sqrt(6.)-2*i,2-i):;
affichage(Q3,3+rempli);
Q5:=triangle(2+i*(sqrt(6.)-4),4-sqrt(6.)+i*(sqrt(6.)-4),2-i):;
affichage(Q5,5+rempli);
Q6:=polygone(2-2*i,2+i*(sqrt(6.)-4),4-sqrt(6.)+i*(sqrt(6.)-4),
             2-2/sqrt(6.)-2*i):;  ;
affichage(Q6,6+rempli);
Q4:=carre(0,1):;
affichage(Q4,4+rempli);
affichage(translation(5/2,Q1),1+rempli);
affichage(translation(1+3*i+5/2,Q5),5+rempli);
```

```
affichage(translation(1+3*i+5/2,Q3),3+rempli);
affichage(translation(5/2,Q2),2+rempli);
affichage(translation(1+3*i+5/2,Q6),6+rempli);
affichage(translation(2-i+5/2,Q4),4+rempli);
affichage(translation(6,Q2),2+rempli);
affichage(translation(2-i+6,Q4),4+rempli);
affichage(translation(1+3*i+6,Q6),6+rempli);
affichage(translation(3-sqrt(6)+sqrt(6)*i+6,Q3),3+rempli);
affichage(translation(2-sqrt(6)+(sqrt(6)-3)*i+6,Q1),1+rempli);
affichage(translation(-1+(3-sqrt(6))*i+6,Q5),5+rempli);
```

On obtient 6 pièces :



11.14 Puzzle transformant une croix de Lorraine en un carrè

11.14.1 Un puzzle en 9 pièces

On tape :

```
L:=0,-i,-2*i,-3*i,1-3*i,1-2*i,1-i,1,2,3,3+i,2+i,1+i,1+2*i;
L:=L,2+2*i,2+3*i,1+3*i,1+4*i,4*i,3*i,-1+3*i,-1+2*i,2*i,i;
L:=L,-1+i,-2+i,-2,-1;
papier_quadrille(1,pi/2,1);
C:=polygone(L):;
affichage(C,1+epaisseur_ligne_2);
carre(1-2*i,4);
```

On obtient :



Comment transformer cette croix en un carré ?
Le polygone C a une aire de 13 carrés de côté 1 unité.
Donc C se transforme en un carré de côté $\sqrt{13}$ unités. $\sqrt{13}$ est l'hypoténuse d'un triangle rectangle de côtés 2 unités et 3 unités. On a 13=1+4*(2*3/2) et donc ce carré est formé par 1 carré de côté 1 unité et par 4 triangles rectangles de côtés 2 unités et 3 unités.

Voici un découpage en 9 pièces.
On tape :

```
affichage(C,1+epaisseur_ligne_1);
carre(1-2*i,4);
R1:=polygone(0,-i/2,1-2*i,1,3,3+i,1+i,1+2*i,2+2*i,
             2+3*i,5/3*i,i,-1+i,-1/3):;
affichage(R1,1+rempli);
R2:=polygone(5/3*i,2+3*i,1+3*i,1+4*i,4*i,3*i):;
affichage(R2,2+rempli);
R3:=carre(-1+2*i,2*i):;
affichage(R3,3+rempli);
R4:=polygone(-i/2,-2*i,1-2*i):;
affichage(R4,4+rempli);
R5:=polygone(-2*i,-3*i,1/2-3*i,1-8*i/3,1-2*i):;
affichage(R5,5+rempli);
R0:=polygone(1/2-3*i,1-3*i,1-8*i/3):;
affichage(R0,80+rempli);
R6:=polygone(-1/3,-1+i,-4/3+i,-4/3):;
affichage(R6,6+rempli);
```

```
R7:=triangle(-2,-4/3,-2+i):;
affichage(R7,47+rempli);
R8:=triangle(-4/3,-4/3+i,-2+i):;
affichage(R8,68+rempli);
```

On obtient 9 pièces :



On tape :

```
affichage(R0,80+rempli);
affichage(R1,1+rempli);
affichage(R2,2+rempli);
affichage(R3,3+rempli);
affichage(R4,4+rempli);
affichage(R5,5+rempli);
affichage(R6,6+rempli);
affichage(R7,47+rempli);
affichage(R8,68+rempli);
carre(5-i,8+i);
affichage(translation(4+i,R1),1+rempli);
affichage(translation(6-2*i,R3),3+rempli);
affichage(translation(6+3*i,R5),5+rempli);
affichage(translation(13/3+i,R6),6+rempli);
affichage(translation(5+2*i,rotation(-2*i,-pi/2,R4)),4+rempli);
affichage(translation(5-2*i,rotation(4*i,pi/2,R2)),2+rempli);
affichage(translation(3+4*i,rotation(1-3*i,pi/2,R0)),80+rempli);
affichage(translation(6+2*i,rotation(-2,pi/2,R7)),47+rempli);
affichage(translation(25/3,rotation(-4/3+i,pi/2,R8)),68+rempli);
```

On obtient la croix et le carré formés avec ces 9 pièces :



11.14.2 Un autre découpage en 9 pièces

Voici un autre découpage en 9 pièces.
On tape :

```
affichage(C,1+epaisseur_ligne_1);
carre(5-i,8+i);//carre(1-2*i,4);
V1:=polygone(0,-i/2,1-2*i,1,3,3+i,1+i,1+2*i,2+2*i,
             2+3*i,5/3*i,i,-1+i,-1/3):;
affichage(V1,1+rempli);
V2:=polygone(5/3*i,2+3*i,-1+3*i,-1+2*i,2*i):;
affichage(V2,2+rempli);
V3:=carre(3*i,1+3*i):;
affichage(V3,3+rempli);
V4:=polygone(-i/2,-2*i,1-2*i):;
affichage(V4,4+rempli);
V5:=polygone(-2*i,-3*i,1/2-3*i,1-8*i/3,1-2*i):;
affichage(V5,5+rempli);
V0:=polygone(1/2-3*i,1-3*i,1-8*i/3):;
affichage(V0,80+rempli);
V6:=polygone(-1/3,-1+i,-4/3+i,-4/3):;
affichage(V6,6+rempli);
V7:=triangle(-2,-4/3,-2+i):;
affichage(V7,47+rempli);
V8:=triangle(-4/3,-4/3+i,-2+i):;
affichage(V8,68+rempli);
```

```
affichage(translation(4+i,V1),1+rempli);
affichage(translation(6-2*i,V2),2+rempli);
affichage(translation(5-i,V3),3+rempli);
affichage(translation(7+3*i,V4),4+rempli);
affichage(translation(6+4*i,rotation(-2*i,pi/2,V5)),5+rempli);
affichage(translation(7-4*i/3,rotation(-1+i,-pi/2,V6)),6+rempli);
affichage(translation(3+4*i,rotation(1-3*i,pi/2,V0)),80+rempli);
affichage(translation(6+2*i,rotation(-2,pi/2,V7)),47+rempli);
affichage(translation(22/3+2*i,rotation(-4/3+i,-pi,V8)),68+rempli);
```

On obtient 9 pièces ou 12 pièces si on coupe la pièce rouge en 4 selon les traits noirs correspondant aux segments :

```
segment(i,1+i);
segment(4+2*i,5+2*i);
segment(1,1+i);
segment(5+i,5+2*i);
segment(2,2+i);
segment(6+i,6+2*i);
```



Cette décompostion en 12 pièces met en évidence que le carré est composé de les 4 triangles rectangles 2*3 qui entourent le carré central unité.

11.14.3 Un puzzle en 8 pièces

Pour avoir les pièces et leurs affichages, on tape :

```
H1:=polygone(0,1,1-i,2-i,2-2*i,4/3-2*i):;
affichage(translation(-2+4*i,H1),1+rempli);
```

```
H2:=polygone(0,1,1-i,2/3-i):;
affichage(translation(-2+13*i/6,H2),2+rempli);
H3:=polygone(0,1/3,i/2):;
affichage(translation(-3/2+2*i/3,H3),3+rempli);
H4:=polygone(0,5/3,1+i,i/3):;
affichage(translation(2+3*i,H4),4+rempli);
H5:=polygone(0,i,-1+i,-1+2*i,2*i,3*i,4/3+i,1+i,1,2,1-2*i/3,-4*i/3,-i,
affichage(translation(7/6+i,H5),5+rempli);
H6:=polygone(0,1,1+3*i,7*i/3):;
affichage(translation(10/3,H6),6+rempli);
affichage(translation(-3/2,H3),68+rempli);
H7:=polygone(0,1,1+i/3,1/2+i/3):;
affichage(translation(-1/2+4*i/3,H7),47+rempli);
```

On obtient :



Exercice
Avec ces pièces : faire la croix de lorraine, puis faire un carré.

Pour faire le puzzle de la croix et du carré, on tape :

```
affichage(H5,5+rempli);
affichage(translation(3*i,H1),1+rempli);
affichage(translation(-11/3*i,H6),6+rempli);
affichage(translation(-4*i,H7),47+rempli);
affichage(translation(-11*i/3,rotation(0,-pi/2,H3)),68+rempli);
affichage(translation(-2-i,H3),3+rempli);
```

```
affichage(translation(1-i,H4),4+rempli);
affichage(translation(2,H2),2+rempli);
affichage(translation(6+3*i,rotation(0,-pi/2,H1)),1+rempli);
affichage(translation(6,H5),5+rempli);
affichage(translation(6,rotation(0,pi/2,H6)),6+rempli);
affichage(translation(4,rotation(0,pi,H3)),3+rempli);
affichage(translation(3+i,H7),47+rempli);
affichage(translation(4+4*i/3,rotation(0,pi/2,H3)),68+rempli);
affichage(translation(8,rotation(0,pi,H2)),2+rempli);
affichage(translation(6-i,rotation(0,pi,H4)),4+rempli);
```

On obtient :



11.14.4 Un autre puzzle en 8 pièces

Pour avoir les pièces, on tape :

```
S1:=triangle(0,sqrt(13.),i):;
S2:=triangle(0,13-3*sqrt(13.),(sqrt(13.)-3)*i):;
S3:=polygone(0,-5+2*sqrt(13.),-5+2*sqrt(13.)+i*(sqrt(13.)-3),
              8-2*sqrt(13.)+i,i):;
S4:=polygone(0,3,3+i,2+i,2+2*i,1+2*i,1+i,i):;
S5:=polygone(0,sqrt(13.)-1,sqrt(13.)-1+i,i):;
S6:=carre(0,1):;
S7:=polygone(0,4-sqrt(13.),4-sqrt(13.)+i,i):;
S8:=polygone(0,sqrt(13.)-3,sqrt(13.)-3+i,i):;
```

Pour les afficher, on tape :

```
affichage(translation(-6+3*i,S1),1+rempli);
```

```
affichage(translation(-2+3.5*i,S2),2+rempli);
affichage(translation(-6-i,S3),3+rempli);
affichage(translation(-3+i,S4),4+rempli);
affichage(translation(-6+i,S5),5+rempli);
affichage(translation(-1-i,S6),6+rempli);
affichage(translation(-2-i,S7),47+rempli);
affichage(translation(-3-i,S8),68+rempli);
```

Voici l'affichage des pièces :



Ces pièces sont formées en transformant le rectangle 1*13 et un carré de côté $\sqrt{13}$ selon la méthde de la section 11.9.

Exercice
Avec ces pièces : faire la croix de lorraine, puis faire un carré.

Pour faire le puzzle de la croix et du carré, on tape :

```
S3:=polygone(0,-5+2*sqrt(13.),-5+2*sqrt(13.)+i*(sqrt(13.)-3),
             8-2*sqrt(13.)+i,i):;
S8:=polygone(0,sqrt(13.)-3,sqrt(13.)-3+i,i):;
affichage(S4,4+rempli);
affichage(translation(1-i,S6),6+rempli);
affichage(translation(3*sqrt(13)-9-2*i,S2),2+rempli);
affichage(translation(-4+sqrt(13)-2*i,S3),3+rempli);
affichage(translation(-1-2*i,S8),68+rempli);
affichage(translation(4-i,rotation(0,pi,S1)),1+rempli);
affichage(translation(1-2*i,rotation(0,-pi/2,S7)),47+rempli);
```

```
affichage(translation(11/2-2*i,S5),5+rempli);
affichage(translation(9/2+i,rotation(0,-pi/2,S4)),4+rempli);
affichage(translation(4-sqrt(13)+5+1/2,S3),3+rempli);
affichage(translation(4+1/2+i,S2),2+rempli);
affichage(translation(sqrt(13)+4+1/2+(sqrt(13)-2)*i,
          rotation(0,pi,S1)),1+rempli);
affichage(translation(6+1/2-i,S6),6+rempli);
affichage(translation(7+1/2-i,S8),68+rempli);
affichage(translation(5+1/2,S7),47+rempli);
affichage(translation(3*sqrt(13)-9-2*i,S2),2+rempli);
affichage(translation(2-5*i,rotation(0,pi/2,S5)),5+rempli);
```

On obtient :



11.15 La quadrature d'un triangle équilatéral ou puzzle de Dudeney

11.15.1 Le carré, le rectangle $1 * \sqrt{3}$ et le triangle équilatéral avec 8 et 6 pièces

On peut transformer un rectangle de dimension $a * b$ en un triangle isocèle à l'aide de 4 pièces : 2 triangles rectangles et 2 trapèzes rectangles.
Le triangle est obtenu en faisant subir au triangle LDU (resp NUA) une rotation de centre L (resp N) et d'angle π : le triangle a comme base de longueur $2 * a$ et la hauteur reliée à cette base a pour longueur b.



Remarques On peut aussi de la même façon transformer un rectangle de dimension $a * b$ en un triangle isocèle de base $2 * b$ et de hauteur a relative à cette base.
On remarquera aussi qu'il suffit de 3 pièces pour réaliser cette transformation (le partage rouge/rose et bleu/jaune etant inutile !).
À condition que les angles de la base b soient aigus, tout triangle de base b et de hauteur $2 * a$ relative à cette base se transforme de cette façon en un rectangle $a * b$.

On va donc transformer selon cette mèthode un rectangle $1 * \sqrt{3}$ en un triangle équilatéral et en un carré.
On tape :

```
b:=sqrt(3);a:=1;
A:=point(0);
B:=point(b);
C:=point(b+i*a);
D:=point(i*a);
rectangle(A,B,C,D);
P:=point(sqrt(a*b));
M:=point(b-sqrt(a*b)+i*a);
Q:=point(sqrt(a*b)*(1+i));
R:=point(sqrt(a*b)*i);
S:=point(sqrt(a*b)+i*a);
T:=point(sqrt(a*b)+i*(sqrt(a*b)-a));
carre(A,P,Q,R, affichage=rouge);
segment(P,D);
segment(B,M);
segment(R,M,affichage=bleu);
segment(P,S, affichage=vert);
U:=point(a*i/2);
V:=point(a*i/2+b-sqrt(b)/2);
W:=point(b+a*i/2);
segment(U,W);
segment(U,L);
segment(U,N);
N:=point(b/2);
O:=point(b/2+i*a/2);
L:=point(b/2+i*a);
```

```
F:=inter_unique(droite(U,L),droite(M,B));
```



Puis, on tape :

```
c:=3/2;
T0:=triangle(A,N,U,affichage=0+rempli);
T1:=polygone(N,P,T,V,U,affichage=1+rempli);
T2:=triangle(T,P,B,affichage=2+rempli);
T3:=triangle(B,W,V,affichage=3+rempli);
T4:=polygone(C,S,L,F,V,W,affichage=4+rempli);
T5:=triangle(U,V,F,affichage=5+rempli);
T6:=triangle(F,L,M,affichage=6+rempli);
T7:=polygone(F,M,D,U,affichage=47+rempli);
affichage(translation(-c*i,T1),1+rempli);
affichage(translation(-sqrt(a*b)-c*i+a*i,T2),2+rempli);
affichage(translation(sqrt(a*b)-b-c*i+(sqrt(a*b)-a)*i,T4),
           4+rempli);
affichage(translation(sqrt(a*b)-b-c*i+(sqrt(a*b)-a)*i,T6),
           6+rempli);
affichage(translation(sqrt(a*b)-b-c*i+(sqrt(a*b)-a)*i,T3),
          3+rempli);
affichage(translation(-c*i,T1),1+rempli);
affichage(translation(-c*i,T0),0+rempli);
affichage(translation(-c*i,T7),47+rempli);
affichage(translation(-c*i,T5),5+rempli);
affichage(translation(3/2-3*c/4*i,T1),1+rempli);
affichage(rotation(N+3/2-3*c/4*i,pi,translation(3/2-3*c/4*i,T0)),
          0+rempli);
affichage(translation(3/2-3*c/4*i,T2),2+rempli);
affichage(translation(3/2-3*c/4*i,T3),3+rempli);
affichage(translation(3/2-3*c/4*i,T4),4+rempli);
```

```
affichage(translation(3/2-3*c/4*i,T5),5+rempli);
affichage(rotation(L+3/2-3*c/4*i,pi,
           translation(3/2-3*c/4*i,T6)),6+rempli);
affichage(rotation(L+3/2-3*c/4*i,pi,
           translation(3/2-3*c/4*i,T7)),47+rempli);
```

On obtient donc 8 pièces :



On obtient 6 pièces si on ne fait pas le partage UW :



11.15.2 Un autre découpage du rectangle $1 * \sqrt{3}$ avec 6 pièces

On tape :

```
T3:=triangle(sqrt(3),sqrt(sqrt(3)),sqrt(sqrt(3))+i*(sqrt(sqrt(3))-1)):;
P:=inter_unique(droite(y=sqrt(3)/3*x),y=-sqrt(sqrt(3)/3)*(x-sqrt(3)));
T1:=polygone(0,sqrt(sqrt(3)),sqrt(sqrt(3))+i*(sqrt(sqrt(3))-1),P):;
T2:=triangle(sqrt(3),P,sqrt(3)+i):;
T4:=triangle(sqrt(3)+i,P,i+sqrt(3)-sqrt(sqrt(3))):;
T5:=polygone(i,i/2+sqrt(3)/2,P,i+sqrt(3)-sqrt(sqrt(3))):;
T6:=triangle(i,0,i/2+sqrt(3)/2):;
affichage(T1,1+rempli);
affichage(T6,6+rempli);
affichage(T2,2+rempli);
affichage(T3,3+rempli);
affichage(T4,4+rempli);
affichage(T5,5+rempli);
affichage(translation(-2*i+i/2,T1),1+rempli);
affichage(translation(-2*i+i/2,T6),6+rempli);
affichage(translation(-2*i+i/2,T5),5+rempli);
affichage(translation(-3*i-sqrt(3)+sqrt(sqrt(3))+i*sqrt(sqrt(3))+i/2,T2),
            2+rempli);
affichage(translation(-3*i-sqrt(3)+sqrt(sqrt(3))+i*sqrt(sqrt(3))+i/2,T4),
            4+rempli);
affichage(translation(-i-sqrt(sqrt(3))+i/2,T3),3+rempli);
affichage(translation(-i/2+3/2,T1),1+rempli);
affichage(translation(-i/2+3/2,T3),3+rempli);
```

```
affichage(translation(-i/2+3/2,T2),2+rempli);
affichage(translation(-3*i/2+3/2,T4),4+rempli);
affichage(translation(-3*i/2+3/2,T5),5+rempli);
affichage(translation(-3/2*i+3/2+sqrt(3),rotation(0,pi/3,T6)),6+rempli
```

On obtient 6 pièces :



11.15.3 Le carré, le rectangle et le triangle équilatéral avec 5 pièces

Voici les 5 pièces d'un puzzle qui peuvent s'assembler soit en un carré soit en un rectangle, soit en 2 paralélogramme, soit en un triangle équilatéral :



On choisit au départ un rectangle de côtés de longueur :
$\sqrt{6}$ et $b = \sqrt{6} * sqrt3/4$.
Ce rectangle a la même surface qu'un triangle équilatéral de côté $\sqrt{6}$ et a aussi la même surface qu'un carré de côté $a = \sqrt{3 * \sqrt{3}/2}$.
On tape :

```
a:=sqrt(3*sqrt(3)/2);
b:=sqrt(3)/4*sqrt(6);
rectangle(0,sqrt(6),sqrt(3)/4);
A:=point(i*b)+sqrt(6)/4);
B:=point(i*b+3*sqrt(6)/4);
T1:=triangle(0,A,i*b);
T2:=triangle(sqrt(6),B,i*b+sqrt(6));
T3:=triangle(sqrt(6),B,sqrt(6)-a+i*b);
T4:=triangle(a,sqrt(6),a+i*(a-b));
T5:=polygone(A,0,a,a+i*(a-b),i*b+sqrt(6)-a);
```

On obtient :



On remarque qu'il suffit de faire :

```
rotation(A,-pi,T1);
rotation(B,pi,T2);
```

pour obtenir un triangle équilatéral. Et on sait facilement transformer ce rectandle en carré avec les 3 pièces : (T1+ T5),T4, (T2+T3) selon la méthode de la section 11.9.
On tape :

```
//le rectangle
affichage(T1,1+rempli);
```

```
affichage(T2,2+rempli);
affichage(T3,3+rempli);
affichage(T4,4+rempli);
affichage(T5,5+rempli);
//le triangle equilateral
affichage(translation(2*i,T5),5+rempli);
affichage(translation(2*i,T4),4+rempli);
affichage(translation(2*i,T3),3+rempli);
affichage(translation(2*i,rotation(B,pi,T2)),2+rempli);
affichage(translation(2*i,rotation(A,-pi,T1)),1+rempli);
//le carre
affichage(translation(2*i+3,T5),5+rempli);
affichage(translation(2*i+3,T1),1+rempli);
affichage(translation(2*i+i*(a-b)+3+a-sqrt(6),T2),2+rempli);
affichage(translation(2*i+i*(a-b)+3+a-sqrt(6),T3),3+rempli);
affichage(translation(2*i+i*b+3-a,T4),4+rempli);
//le parallelogramme de base sqrt(6) et de hauteur b
affichage(translation(3.5,T5),5+rempli);
affichage(translation(3.5,T4),4+rempli);
affichage(translation(3.5,T3),3+rempli);
affichage(translation(3.5,T2),2+rempli);
affichage(translation(3.5+sqrt(6),T1),1+rempli);
//l'autre parallelogramme de base a et de hauteur a
affichage(translation(2.5*i+5,T5),5+rempli);
affichage(translation(2.5*i+i*b+5-a,T4),4+rempli);
affichage(translation(2.5*i+5,T1),1+rempli);
affichage(translation(2.5*i-i*b+5+a-sqrt(6),T2),2+rempli);
affichage(translation(2.5*i-i*b+5+a-sqrt(6),T3),3+rempli);
```

On obtient les 5 quadilatères du début :



11.15.4 Le puzzle de Dudeney : carré ou triangle équilatéral avec 4 pièces

Voici les 4 pièces du puzzle de Henry Ernest Dudeney (1857-1930) qui peuvent s'assembler soit en un carré soit en un triangle équilatéral :



Etant donné un carré, trouver la construction de ces pièces et prouvez que l'on a bien obtenu après réorganisation un triangle équilatéral.

La solution



Si le carré est de côté `2a` et le côté du triangle équilatéral est de côté `2b`.
On doit avoir (égalité des aires) :
`4a^2=b^2*sqrt(3)` On pose `l:=sqrt(b^2-a^2)`.
On voit que si les figures sont exactes pour obtenir le triangle à partir du carré de côté `2a` il faut, sans bouger la pièce bleue, faire subir :

1. à la pièce rouge une symétrie par rapport à `M:=point(1)`
2. à la pièce jaune une symétrie par rapport à `Q:=point(2+i*l)`
3. à la pièce verte une translation de vecteur `PP1` si `P1` est le symétrique de `P:=point((a-l)*i)` par rapport à `M`.

On doit donc avoir :
le côté du triangle équilatéral $2b$ est égal à $PR + PN + RN = 2PN = 2b$,
$AM = MB$ donc M est le milieu de AB,
$DN = NC$ donc N est le milieu de CD,
$CQ + DP = AP + QB = 2a$
$RM = RQ = b$
$\overbrace{PRM} = \overbrace{MRQ} = \overbrace{QRN} = \pi/3$
$4a^2 = b^2\sqrt{3}$ car le carré et le triangle ont même surface.
Donc RMQ est équilatéral puisque c'est un triangle isocèle ayant un angle de $\pi/3$ et donc $\overbrace{RMQ} = \overbrace{PRM} = \overbrace{QRN} = \pi/3$.
Puisque on a aussi $\overbrace{MQR} = \overbrace{QRN} = \pi/3$, on en déduit que :
MQ est parallèle à PN et NQ est parallèle à PM.
les triangles DNP et BMQ sont égaux.
les triangles AMP et CNQ sont égaux.
le quadrilatère $PMQN$ est un parallélogramme.
On a donc $AP = QC = 2a - l$ et $PD = BQ = l$ et
aire($PMQN$)=$4a^2 - al - (2a - l)a = 2a^2 = h * \sqrt{a^2 + l^2} = hb$ si h est la

distance entre les parallèles PN et MQ.
Donc $hb = 2a^2 = b^2\sqrt{3}/2$ et on en déduit $h = b\sqrt{3}/2$.
Réciproquement, si $PD = BQ = l$ et si RMQ est équilatéral alors :
MQ est parallèle à PN et R se trouve sur PN car la hauteur du triangle RMQ qui vaut $b\sqrt(3)/2$ est aussi égale à la distance h entre les parallèles PN et MQ.
$\widehat{PRM} = \widehat{MRQ} = \widehat{QRN} = \pi/3$
Donc $\widehat{PRM} = \widehat{MQR} = \pi/3$ et $\widehat{QRN} = \widehat{MQR} = \pi/3$
Autre démonstration Si le carré a comme côté $2a$ et le triangle équilatéral a comme côté $2b$ on a :
$4a^2 = b^2\sqrt{3}$ car le carré et le triangle ont même surface.
Posons $l = PD$ on a $b^2 = PN^2 = l^2 + a^2 = 4a^2/\sqrt(3)$
Donc $l^2 = a^2(4\sqrt{3}/3 - 1)$
Si $PD = BQ = l$ et $AM = MB = DN = NC = a$ on a :
$PN = MQ$ et $PM = NQ$ donc la quadrilatère $MPNQ$ est un parallélogramme.
NM est paralléle à DP donc $\widehat{NPD} = \widehat{PNM} = \alpha$. $\sin(\widehat{NPD}) = a/b$ et $\sin(\widehat{PNM}) = h/2a$ donc $h = \frac{2a^2}{b} = \frac{b\sqrt{3}}{2}$.
Le point R est donc l'intersection de la médiatrice de MQ avec PN. Le triangle PQR est donc isocèle est sa hauteur issue de R vaut $h = b\frac{b\sqrt{3}}{2}$ donc le triangle PQR est équilatéral.
Comment construire le point R ?
On a montré que R est le transformé de Q dans la rotation de centre M et d'angle $\pi/3$.
On tape pour avoir les pièces du puzzle (on prend pour simplifier $a = 1$) i.e. le carré a des côtés de longueur 2 :

```
l:=sqrt(4*sqrt(3)/3-1);
M:=point(1):;
P:=point(i*(2-l)):;
Q:=point(2+i*l):;
R:=rotation(M,pi/3,Q):;
quadrilatere(M,2,Q,R,affichage=4+rempli);
quadrilatere(Q,2+2*i,1+2i,R,affichage=3+rempli);
triangle(P,2i,1+2i,affichage=2+rempli);
quadrilatere(0,M,R,i*(2-l),affichage=1+rempli);
```

On obtient :



Comment construire les pièces à partir d'un triangle équilatèral
Soit un triangle de côté $2b = 2$.
On tape :

```
triangle_equilateral(0,2,R));
b:=1;
a2:=b^2*sqrt(3)/4;
l2:=1-a2;
M:=milieu(0,R);
Q:=milieu(2,R);
B:=inter_unique(cercle(M,sqrt(a2)),cercle(Q,sqrt(l2)),point(2));
T4:=polygone(R,M,B,Q);
P:=inter_unique(droite(Q,B),droite(0,2));
T1:=polygone(0,P,B,M);
S:=P+1;
N:=inter_unique(droite(B,Q),perpendiculaire(S,droite(B,Q)));
T3:=polygone(2,Q,N,S);
T2:=triangle(N,P,S);
```

On obtient :



Pour animer la figure
On peut animer la figure, soit manuellement avec les curseurs, soit avec la commande `animation`.
On tape :

```
animtri(t1,t2,t3):={
  local L,l,M,P,P1,Q,R;
l:=sqrt(4*sqrt(3.)/3-1);
M:=point(1):;
P:=point(i*(2-l)):;
P1:=symetrie(M,P):;
Q:=point(2+i*l):;
R:=rotation(M,pi/3,Q):;
L:=quadrilatere(M,2,Q,R,affichage=4+rempli);
L:=L,affichage(rotation(M,t1,quadrilatere(0,M,R,i*(2-l))),
               1+rempli);
L:=L,affichage(rotation(Q,-t3,quadrilatere(Q,2+2*i,1+2i,R)),
               3+rempli);
L:=L,affichage(translation((P1-P)*t2,triangle(i*(2-l),2i,1+2i)),
               2+rempli);
return L;
}:;
```

Puis on utilise des paramètres dans un écran de géométrie, on tape :

```
supposons(t1=[0.0,0,3.14,0.1]);
supposons(t3=[0.0,0,3.14,0.1]);
supposons(t2=[0.0,0,1,0.1]);
animtri(t1,t2,t3);
```

En faisant bouger les curseurs `t1`,`t3` et `t2`, on voit les pièces du puzzle se déplacer selon les transformations annoncées.
Ou bien :
On crée une animation pour voir les déplacements, pour cela on crée les listes que la commande `animation` va utiliser.
On tape :

```
T1:=seq([animtri(t1,0,0)],t1=0..3.14,3.14/10):;
T3:=seq([animtri(3.14,0,t3)],t3=0..3.14,3.14/10):;
T2:=seq([animtri(3.14,t2,3.14)],t2=0..1,0.1):;
T4:=seq([animtri(t1,1,3.14)],t1=3.14..0,3.14/10):;
T6:=seq([animtri(0,1,t3)],t3=3.14..0,3.14/10):;
T5:=seq([animtri(0,t2,0)],t2=1..0,0.1):;
```

Puis on tape :
`animation(T1,T3,T2,T4,T6,T5)`
On obtient :



11.15.5 Une autre animation

```
animtri1(t2,t3,t4):={
local l,M,P,Q,R,Q1,Q2,P3,P2,L;
l:=sqrt(4*sqrt(3.)/3-1);
M:=point(1):;
P:=point(i*(2-l));
Q:=point(2+i*l);
R:=rotation(M,pi/3,Q):;
L:=quadrilatere(M,0,P,R,affichage=1+rempli);
L:=L,affichage(rotation(M,t4,quadrilatere(M,2,Q,R)),4+rempli);
P3:=rotation(M,t4,quadrilatere(Q,R,1+2*i,2+2*i));
Q1:=symetrie(M,Q);
P2:=rotation(Q1,t3,rotation(M,t4,triangle(i*(2-l),2i,1+2i)));
L:=L,affichage(rotation(Q1,t3,P3),3+rempli);
Q2:=rotation(Q1,t3,symetrie(M,1+2*i));
L:=L,affichage(rotation(Q2,t2,P2),2+rempli);
return L;
 }:;
T14:=seq([animtri1(0,0,t4)],t4=0..-3.14,-3.14/10):;
T13:=seq([animtri1(0,t3,-3.14)],t3=0..-3.14,-3.14/10):;
T12:=seq([animtri1(t2,-3.14,-3.14)],t2=0..-3.14,-3.14/10):;
T15:=seq([animtri1(t2,-3.14,-3.14)],t2=-3.14..0,3.14/10):;
T16:=seq([animtri1(0,t3,-3.14)],t3=-3.14..0,3.14/10):;
T10:=seq([animtri1(0,0,t4)],t4=-3.14..0,3.14/10):;
```

Puis on tape :
`animation(T14,T13,T12,T15,T16,T10)`
On obtient :



11.15.6 Encore une autre animation

```
animtri2(a):={
local l,M,N,B,C,D1,P,Q,R,Q1,N1,B1,C1,N2,P3,P2,L,B,C,R1,P1;
l:=sqrt(4*sqrt(3.)/3-1);
B:=point(2);C:=point(2+2*i);
M:=point(1);N:=point(1+2*i);
P:=point(i*(2-l));
Q:=evalf(point(2+i*l));
R:=evalf(rotation(M,pi/3.,Q));
L:=quadrilatere(M,0,P,R,affichage=1+rempli);
L:=L,affichage(rotation(M,a,evalf(quadrilatere(M,2,Q,R))),4+rempli);
Q1:=rotation(M,a,Q);
R1:=rotation(M,a,R);
N1:=rotation(M,a,N);
B1:=rotation(M,a,B);
C1:=rotation(M,a,C);
D1:=rotation(Q1,a,rotation(M,a,point(2*i)));
P1:=rotation(Q1,a,rotation(M,a,P));
N2:=rotation(Q1,a,N1);
P3:=quadrilatere(Q1,R1,N1,C1);
L:=L,affichage(rotation(Q1,a,P3),3+rempli);
P2:=triangle(P1,D1,N2);
L:=L,affichage(rotation(N2,a,P2),2+rempli);
return L;
 }:;
T24:=seq([animtri2(t4)],t4=0..-3.14,-3.14/10):;
T20:=seq([animtri2(t4)],t4=-3.14..0,3.14/10):;
```

Puis on tape :
`animation(T24,T20)`
On obtient :



11.15.7 Le pavage correspondant au puzzle de Dudeney

On tape les procédures `dudeney0` (la pièce triangle est en bas à gauche) et `dudeney1` (la pièce triangle est en haut à droite) :

```
dudeney0(z0):={
local M,P,Q,R,l,L;
l:=sqrt(4*sqrt(3)/3-1);
M:=point(1+z0):;
P:=point(2+i*(2-l)+z0):;
Q:=point(i*l+z0):;
L:=NULL;
R:=rotation(M,-pi/3,Q):;
L:=L,quadrilatere(M,z0,Q,R,affichage=4+rempli);
L:=L,quadrilatere(Q,2*i+z0,1+2i+z0,R,affichage=3+rempli);
L:=L,triangle(P,2+2i+z0,1+2i+z0,affichage=2+rempli);
L:=L,quadrilatere(2+z0,M,R,i*(2-l)+z0+2,affichage=1+rempli);
retourne L;
}:;
dudeney1(z0):={
local M,P,Q,R,l,L;
l:=sqrt(4*sqrt(3)/3-1);
M:=point(1+z0+2*i):;
P:=point(i*(l)+z0):;
Q:=point(2-i*l+2*i+z0):;
L:=NULL;
R:=rotation(M,-pi/3,Q):;
L:=L,quadrilatere(M,2+z0+2*i,Q,R,affichage=4+rempli);
L:=L,quadrilatere(Q,2+z0,1+z0,R,affichage=3+rempli);
L:=L,triangle(P,z0,1+z0,affichage=2+rempli);
L:=L,quadrilatere(0+z0+2*i,M,R,i*(l)+z0,affichage=1+rempli);
retourne L;
}:;
```

Puis on tape :

```
dudeney1(0);
dudeney0(-2*i));
```

```
dudeney0(-2+2*i(l-1));
dudeney1(-2-2*i+2*i(l-1));
```

On obtient les 4 carrés et l'apparition d'un triangle équilatéral :



On peut compléter ce dessin pour ontenir un pavage de carrés ou de triangles équilatéraux de même surface.
On tape :

```
dudeney1(0);
dudeney0(-2*i));
dudeney0(-2+2*i(l-1));
dudeney1(-2-2*i+2*i(l-1));
dudeney0(-2-4*i+2*i(l-1));
dudeney1(-4*i));
dudeney0(2+2*i(1-l));
dudeney1(2-2*i*l);
dudeney0(2+2*i*(-1-l));
```

On obtient :



11.16 La quadrature d'un dodécagone

11.16.1 Le dodécagonee

On définit les points utiles.
On tape :

```
LP:=[point(exp(k*i*pi/6))$(k=0..11)]:;
[A,B,C,D,E,F,G,H,I,J,K,L]:=LP;
polygone(LP);
triangle_equilateral(A,B,M);
triangle_equilateral(J,I,N);
triangle_equilateral(F,E,P);
O:=isobarycentre(F,M,I);
```

On obtient :



Exercice
On trace le dodécacone $ABCDEFGHIJKL$ inscrit dans un cercle de rayon 1 unité et le triangle équilatèral ABM comme ci-dessus, puis on trace les 4 segments : cela détermine 6 régions :

```
segment(A,I);
segment(B,F);
segment(A,F);
segment(B,I);
```



Montrer que AF et BI se coupent en M sommet du triangle équilatèral ABM.
Montrer que le triangle FMI est équilatéral.
Calculer les angles des triangles MBF et MAI.
Calculer les angles des triangles AIL et BCF. Montrer que $AI = BF = \sqrt{3}$

11.16.2 Le dodécagone et le carré

Exercice
Soit un dodécaèdre inscrit dans un cercle de rayon 1 unité.
Calculer son aire.
Calculer le côté du carré qui a la même aire que ce dodécagone.
Réponse
Aire du dodécaèdre= $12\sin(2*pi/12)/2 = 6\sin(pi/6) = 3$ unité*unité.
Le côté du carré qui a la même aire est donc : $\sqrt{3}$ unités.
Le découpage
On découpe le dodécaèdre en 6 morceaux, puis on forme un carré avec ces 6 morceaux.
D'après ce qui précède AI et BF ont pour longueur le côté du carré cherché.
On tape :

```
P1:=polygone(A,I,J,K,L):;
affichage(P1,1+rempli);
P2:=polygone(B,C,D,E,F):;
affichage(P2,2+rempli);
P3:=triangle_equilateral(A,B,M):;
affichage(P3,3+rempli);
P4:=triangle(M,I,A):;
affichage(P4,4+rempli);
P5:=triangle(M,F,B):;
```

```
affichage(P5,5+rempli);
P6:=polygone(F,G,H,I,M):;
affichage(P6,6+rempli);
O:=isobarycentre(F,M,I):;
affichage(translation(3,P6),6+rempli);
affichage(translation(3,rotation(O,2*pi/3,P4)),4+rempli)
affichage(translation(3,rotation(O,-2*pi/3,P5)),5+rempli);
affichage(translation(3-A+H,P3),3+rempli);
triangle_equilateral(J,I,N):;
affichage(translation(3,rotation(
                  isobarycentre(I,J,N),pi/2+pi/6,P1)),1+rempli);
triangle_equilateral(F,E,P):;
affichage(translation(3,rotation(
                  isobarycentre(F,E,P),-pi/2-pi/6,P2)),2+rempli);
```

On obtient :



11.16.3 Pour animer la figure

On peut animer la figure, manuellement avec les curseurs.
On tape :

```
affichage(P6,6+rempli);
supposons(t1=[1.0,0,1,0.1]);
affichage(rotation(isobarycentre(I,J,N),-t1*4*pi/3,P1),1+rempli) ;
supposons(t2=[1.0,0,1,0.1]);
affichage(rotation(isobarycentre(F,E,P),(t2)*4*pi/3,P2),2+rempli);
supposons(t3=[1.0,0,1,0.1]);
affichage(translation(t3*(H-A),P3),3+rempli);
supposons(t4=[1.0,0,1,0.1]);
```

```
affichage(rotation(O,t4*2*pi/3,P4),4+rempli);
supposons(t5=[1.0,0,1,0.1]);
affichage(rotation(O,-t5*2*pi/3,P5),5+rempli);
O;
N;
P;
```

On peut aussi animer la figure avec la commande `animation`.
On tape (attention les variables P1,..P6,A,H,E,F,I,J,n,P sont globales) :

```
animdode(t1,t2,t3,t4,t5):={
local L;
L:=NULL;
L:=L,affichage(P6,6+rempli);
L:=L,affichage(rotation(isobarycentre(I,J,N),-t1*4*pi/3.,P1),
                                                    1+rempli);
L:=L,affichage(rotation(isobarycentre(F,E,P),(t2)*4*pi/3.,P2),
                                                    2+rempli);
L:=L,affichage(translation(t3*(H-A),P3),3+rempli);
L:=L,affichage(rotation(O,t4*2*pi/3,P4),4+rempli);
L:=L,affichage(rotation(O,-t5*2*pi/3,P5),5+rempli);
return L;
}:;
```

Puis si on utilise un seul paramètre (tout bouge en même temps ! ! !) :
`K1:=seq([animdode(t1,t1,t1,t1,t1)],t1=0.0..1.0,0.1):;`
`K2:=seq([animdode(t1,t1,t1,t1,t1)],t1=1.0..0.0,0.1):;`
`animation(K1,K2)`
Ou bien si on utilise 5 paramètres :

```
T1:=seq([animdode(t1,0,0,0,0)],t1=0.0..1.0,0.1):;
T2:=seq([animdode(1,t2,0,0,0)],t2=0.0..1.0,0.1):;
T3:=seq([animdode(1,1,t3,0,0)],t3=0.0..1.0,0.1):;
T4:=seq([animdode(1,1,1,t4,0)],t4=0.0..1.0,0.1):;
T5:=seq([animdode(1,1,1,1,t5)],t5=0.0..1.0,0.1):;
T6:=seq([animdode(1,1,1,1,t5)],t5=1.0..0.0,0.1):;
T7:=seq([animdode(1,1,1,t4,0)],t4=1.0..0.0,0.1):;
T8:=seq([animdode(1,1,t3,0,0)],t3=1.0..0.0,0.1):;
T9:=seq([animdode(1,t2,0,0,0)],t2=1.0..0.0,0.1):;
T0:=seq([animdode(t1,0,0,0,0)],t1=1.0..0.0,0.1):;
```

Puis on tape :
`animation(T1,T2,T3,T4,T5,T6,T7,T8,T9,T0)`
Ou encore si on n'utilise que 4 paramètres (en échangeant la pièce bleue avec la pièce violette à l'aide d'un seul paramètre).

```
Q1:=seq([animdode(t1,0,0,0,0)],t1=0.0..1.0,0.1):;
Q2:=seq([animdode(1,t2,0,0,0)],t2=0.0..1.0,0.1):;
Q3:=seq([animdode(1,1,t3,0,0)],t3=0.0..1.0,0.1):;
```

```
Q4:=seq([animdode(1,1,1,t4,t4)],t4=0.0..1.0,0.1):;
Q7:=seq([animdode(1,1,1,t4,t4)],t4=1.0..0.0,0.1):;
Q8:=seq([animdode(1,1,t3,0,0)],t3=1.0..0.0,0.1):;
Q9:=seq([animdode(1,t2,0,0,0)],t2=1.0..0.0,0.1):;
Q0:=seq([animdode(t1,0,0,0,0)],t1=1.0..0.0,0.1):;
```

Puis on tape :
`animation(Q1,Q2,Q3,Q4,Q7,Q8,Q9,Q0)`

11.17 Autre puzzle du dodécahgone et le carré

On découpe le carrè et le dodécagone $ABCDEFGHIJKL$ en 9 pièces :
1 carré, 4 triangles équilatéraux identiques et 4 polygones identiques au polygone $ABCDE$.



Pour cela on définit un dodécagone de centre $z0$ tourné d'un angle de t radians :

```
dode(z0,t):={
  local A,B,C,D,E,F,G,H,I,J,K,L;
  [A,B,C,D,E,F,G,H,I,J,K,L]:=point(exp(i*k*pi/6))$(k=0..11);
  retourne translation(z0,rotation(A,t,polygone(A,B,C,D,E,F,G,H,I,J,K,L)));
      }:;
```

On tape :
`dode(1+i,0),dode(1+i,pi/3)`
On obtient :



Puis on tape :

```
[A,B,C,D,E,F,G,H,I,J,K,L]:=point(exp(i*k*pi/6))$(k=0..11):;
P0:=polygone(A,B,C,D,E,F,G,H,I,J,K,L):;
dode(0,pi/3);
T0:=triangle_equilateral(A,B,S):;
T1:=triangle_equilateral(F,G,M):;
T2:=triangle_equilateral(H,I,N):;
segment(B,F):;
T5:=carre(N,M,P,Q):;
T4:=triangle_equilateral(M,N,M1):;
segment(A,I);
T3:=triangle_equilateral(Q,P,Q1):;
R:=rotation(A,pi/3,F);
isopolygone(R,I,12);
isopolygone(G,F,12);
carre(R,S,S1,R1):;
translation(7/2,carre(R,S)):;
P5:=translation(7/2,T5):;
 affichage(P5,5+rempli);
P1:=translation(7/2,T1):;
affichage(P1,1+rempli);
P2:=translation(7/2,T2):;
affichage(P2,2+rempli);
P3:=translation(7/2,T3):;
affichage(P3,3+rempli);
P4:=translation(7/2,T4):;
affichage(P4,4+rempli);
P6:=translation(7/2,polygone(S,S1,F,M,M1)):;
affichage(P6,6+rempli);
carre(R,S,S1,R1);
P7:=translation(7/2,polygone(R,S,M1,N,I)):;
affichage(P7,47+rempli);
P8:=translation(7/2,polygone(R1,R,I,H,Q1)):;
affichage(P8,68+rempli);
P9:=translation(7/2,polygone(R1,Q1,G,F,S1)):;
affichage(P9,229+rempli);
affichage(T1,1+rempli);
```

```
affichage(T2,2+rempli);
affichage(T0,3+rempli);
affichage(T4,4+rempli);
affichage(T5,5+rempli);
T6:=polygone(B,F,M,M1,S):;
affichage(T6,6+rempli);
T7:=polygone(A,S,M1,N,I):;
affichage(T7,47+rempli);
T8:=polygone(B,C,D,E,F):;
affichage(T8,68+rempli);
T9:=polygone(A,I,J,K,L):;
affichage(T9,229+rempli);
isopolygone(S1,F,12);
```

11.18 L'hexagone, le rectangle et du carré

11.18.1 Premièren façon

On considère un rectangle de côtés 3 et $\sqrt{3}/2$.
On transforme facilement ce rectangle en un hexagone de côté 1 avec 3 pièces.
On sait transformer ce rectangle en un carré avec 3 pièces selon la méthode vue en section 11.9.
Avec 6 pièces on peut passer de l'hexagone au carré via le rectangle.
On tape :

```
a:=sqrt(3*sqrt(3.)/2);
b:=sqrt(3.)/2;
d:=droite(3,3-a+i*b):;
Q:=inter_unique(droite(y=b),d):;
R:=inter_unique(droite(x=a),d):;
S:=inter_unique(droite(y=2*b*(x-2)),d):;
P1:=polygone(R,Q,1/2+i*b,1,a):;
affichage(P1,1+rempli);
P2:=polygone(R,a,2,S):;
affichage(P2,2+rempli);
P3:=polygone(Q,S,5/2+i*b):;
affichage(P3,3+rempli);
P4:=triangle(S,2,3):;
affichage(P4,4+rempli);
P5:=polygone(S,3,3+i*b,5/2+i*b):;
affichage(P5,5+rempli);
P6:=polygone(i*b,0,1,1/2+i*b):;
affichage(P6,6+rempli);
affichage(translation(i+2,P1),1+rempli);
affichage(translation(i+2,P2),2+rempli);
affichage(translation(i+2,P3),3+rempli);
affichage(translation(i+i*b-3/2+2,P4),4+rempli);
affichage(translation(i+i*b-3/2+2,P5),5+rempli);
```

```
affichage(translation(i+i*b+3/2+2,P6),6+rempli);
affichage(translation(i,P1),1+rempli);
affichage(translation(i,P6),6+rempli);
affichage(translation(i+i*b-a,P2),2+rempli);
affichage(translation(i+i*b-a,P4),4+rempli);
affichage(translation(i+i*(a-b)-3+a,P3),3+rempli);
affichage(translation(i+i*(a-b)-3+a,P5),5+rempli);
```

On obtient :



11.18.2 Deuxième façon

On considère un rectangle de côtés 3 et $2 * \sqrt{3}$.
On transforme facilement ce rectangle en un hexagone de côté 2 avec 3 pièces.
On sait transformer ce rectangle en un carré avec 3 pièces selon la méthode vue en section 11.9.
Avec 6 pièces on peut passer de l'hexagone au carré via le rectangle.
La surface de l'hexagone est `S:=4*sqrt(3)/4*6=6*sqrt(3)` donc le carré aura comme côté `b:=sqrt(6*sqrt(3))`
On tape :

```
b:=sqrt(6*sqrt(3));
hexagone(3*i-4+sqrt(3),-4+2*i):;
polygone(0,2*sqrt(3),2*sqrt(3)+3*i,3*i);
segment(2*i,3*i+sqrt(3));
segment(2*i+2*sqrt(3),3*i+sqrt(3));
segment(2*sqrt(3),i*b);
segment(b,b+i*(b-3));
carre(-i*b-i,-i*b+b-i):;;
p1:=triangle(b,2*sqrt(3),b+i*(b-3)):;
```

```
affichage(p1,1+rempli);
p2:=triangle(2*sqrt(3)+2*i,2*sqrt(3)+3*i,sqrt(3)+i*3):;
affichage(p2,2+rempli);
p3:=triangle(sqrt(3)+3*i,2*sqrt(3)-b+3*i,M):;
affichage(p3,3+rempli);
p4:=polygone(2*sqrt(3),2*sqrt(3)+2*i,sqrt(3)+3*i,M):;
affichage(p4,4+rempli);
p5:=polygone(M,2*i,0,b,b+i*(b-3)):;
affichage(p5,5+rempli);
p6:=polygone(M,2*i,3*i,2*sqrt(3)-b+i*3):;
affichage(p6,6+rempli);
t2:=translation(-3*i-i-2*sqrt(3)+b,p2):;
affichage(t2,2+rempli);
t3:=translation(-3*i-i-2*sqrt(3)+b,p3):;
affichage(t3,3+rempli);
t4:=translation(-3*i-i-2*sqrt(3)+b,p4):;
affichage(t4,4+rempli);
t5:=translation(-4*i-i*(b-3),p5):;
affichage(t5,5+rempli);
t6:=translation(-4*i-i*(b-3),p6):;
affichage(t6,6+rempli);
t1:=translation(-b-i-i*(b-3),p1):;
affichage(t1,1+rempli);
h2:=translation(-3*i-4-sqrt(3),p2):;
affichage(h2,2+rempli);
h3:=translation(-3*i-4+sqrt(3),p3):;
affichage(h3,3+rempli);
h6:=translation(-3*i-4+sqrt(3),p6):;
affichage(h6,6+rempli);
h5:=translation(-4,p5):;
affichage(h5,5+rempli);
h4:=translation(-4,p4):;
affichage(h4,4+rempli);
h1:=translation(-4,p1):;
affichage(h1,1+rempli);
```

On obtient :



11.19 L'hexagone et le carré

Voila la solution de Paul-Jean Busshop (1799-1877) pour transformer en 5 pièces un hexagone régulier en carré sans passer par le rectangle.
Soit un hexagone de coé 1. Son aire vaut donc `3*sqrt(3)/2` et le carré a donc comme coté `c:=sqrt(3*sqrt(3)/2)`.
Busshop transforme l'hexagone en un parallélogramme regroupant ses 2 moitiés puis dessine le carré de côté `c` de façon à ce que les 2 droites formées par deux côtés parallèles du carré passe par les sommets de la base du parallélogramme.

On tape pour voir le carré A, B, C, D et l'hexagone :

```
c:=sqrt(3*sqrt(3)/2);;
I:=point(0);;
O:=point(1/2-i*sqrt(3)/2);;
J:=point(2);A:=point(3);;
K:=point(1/2+i*sqrt(3)/2,affichage=quadrant1); ;
P:=point(3/2-i*sqrt(3)/2,affichage=quadrant2);
isopolygone(O,P,6);;
isopolygone(2,3,6);;
C1:=cercle(3/2,3/2):;;
C2:=cercle(A,c):;;
affichage(C2,2);;
B:=inter_unique(C1,C2,i);;
c:=carre(A,B,C,D):;c;
segment(B,I);;
```

```
segment(K,point(7/2+i*sqrt(3)/2));;
segment(A,I);;
E:=inter_unique(droite(B,C),droite(y=sqrt(3)/2),affichage=quadrant3);;
F:=inter_unique(droite(C,D),droite(y=0));;
p1:=parallele(D,droite(y=sqrt(3)*x)):;;
G:=inter_unique(droite(y=0),p1);;
H:=inter_unique(droite(A,B),droite(y=sqrt(3)/2));;
affichage(p1,2);
p2:=parallele(P,droite(A,B)):;;
affichage(p2,2);;
L:=inter_unique(p2,droite(y=0),affichage=quadrant3);
segment(P,D,affichage=2);
D:=affichage(D,quadrant1);
C:=affichage(C,quadrant1);
M:=point(3/2+i*sqrt(3)/2,affichage=quadrant1)
```

On obtient le carré `ABCD` et les points utiles ainsi que leurs constructions (en vert) :



Le triangle rectangle IAB est tel que $IA = 3$ et $AB = l = \sqrt{3\sqrt{3}/2} = IF$ En effet :
$IB^2 = IA^2 - AB^2 = 9 - l^2 = 9 - 3\sqrt{3}/2$
Dans un repère d'origine I et d'axe des x porté par IA la droite IB a pour équation $y = l/IB * x$ donc E (intersection de $y = \sqrt{3}/2$ avec IB) a pour coordonnées : $IB\sqrt{3}/2/l, \sqrt{3}/2$ donc :
$IE^2 = 3/4(1 + IB^2/l^2) = 3/4 * (l^2 + IB^2)/l^2 = 3 * 9/4/l^2 = 3\sqrt{3}/2 = l^2$
Les triangles IEK et DAG sont donc égaux donc $GA = KF$ (seront en jaune).
Puisque $IB = IC + CB = IC + l = *IE + EB = l + EB$ on a $IC = EB$donc les triangles rectangles IFC et EHB sont égaux donc $IF = FH$ (seront en magenta).
Le quadrilatères $PDGJ$ et $PDFL$ sont des parallélogrammes ayant en commun le côté PD) Donc la translation $\overrightarrow{GJ}$ ($\overrightarrow{FL} = \overrightarrow{GJ} = \overrightarrow{DP}$).

transforme le triangle DGF en le triangle PJL ces triangles sont donc égaux (seront en bleu).
Bien voir que PK passe par le centre de l'hexagone mais le milieu de LF n'est pas ce centre et donc la droite CD ne passe pas par K ! Les quadrilatères $OPLI$ et $JAHM$ (ce ne sont pas des parallélogrammes !) se déduisent par la translation de vecteur $\overrightarrow{OJ}$ (seront en rouge) Le quadrilatère $MECFJ$ est commun à l'hexagone et au carré (sera en vert).



Avec Xcas, on peut calculer différentes longueurs .
On tape :
`simplify(longueur2(P,D)-longueur2(G,J));`
`simplify(longueur2(P,D)-longueur2(L,F))`
On obtient : `(0,0)`
On tape :
`simplify(longueur2(K,E)-longueur2(G,A))`
On obtient :
`0`
On tape :
`simplify(longueur2(I,L)-longueur2(M,H))`
On obtient :
`0`
On tape :
`est_element(K,droite(C,D))`
On obtient :
`0`
On tape :
`simplify(affixe(milieu(L,F)))`
On obtient :
`(99-sqrt(33)*sqrt(37+sqrt(3)*30))/44`$\simeq$ 1.01858227335
On tape :
`longueur2(I,C),longueur2(E,B)`

On obtient après simplification :
`3*sqrt(sqrt(3)*6-3)+9,3*sqrt(sqrt(3)*6-3)+9`
On tape :
`longueur2(J,A)`
On obtient après simplification :
`1`
On tape :
`longueur2(D,A),longueur2(E,I)`
On obtient après simplification :
`3*sqrt(3)/2,3*sqrt(3)/2`
On tape :
`longueur2(G,A),longueur2(E,K)`
On obtient après simplification :
`expr("rootof([[-2,6,-3,93],[1,0,0,-72,-72]])",0)/30,expr("rootof([[-2,6,-3,93`

On définit les pièces du puzzle, on tape :

```
P1:=polygone(A,H,1+i*sqrt(3.),3/2+i*sqrt(3)/2):;
affichage(P1,1+rempli);
P2:=polygone(3/2+i*sqrt(3)/2,1+i*sqrt(3),E,C,F):;
affichage(P2,2+rempli);
P3:=polygone(A,G,D):;
affichage(P3,3+rempli);
P4:=polygone(D,G,F):;
affichage(P4,4+rempli);
P5:=polygone(B,E,H):;
affichage(P5,5+rempli);
isopolygone(3,4,6):;
affichage(translation(-3/2-i*sqrt(3.)/2+3,evalf(P1)),1+rempli);
affichage(translation(3,evalf(P2)),2+rempli);
affichage(translation(evalf(3+i*sqrt(3)-affixe(G)),evalf(P3)),3+rempli);
affichage(translation(evalf(point(4)-D),evalf(P4)),4+rempli);
affichage(translation(point(5/2+i*sqrt(3)/2)-E,evalf(P5)),5+rempli);
```

On obtient :



11.20 L'octogone, le rectangle et le carré

Voici les pièces 8 pièces du puzzle :
On tape pour définir les points utilisés :

```
a:=evalf(2^(3/4));
[A,B,C,D,E,F,G,H]:=point(exp(i*k*pi/4))$(k=0..7);
polygone(A,B,C,D,E,F,G,H);
P1:=polygone(A,B,D,E,F,H):;P1;
I:=milieu(B,D);
J:=milieu(F,H);
[A1,B1,D1,E1,F1,H1]:=translation(5/2,[A,B,D,E,F,H]);
T1:=translation(5/2,P1):;T1;
P2:=triangle(B,C,I):;P2;
P3:=triangle(C,D,I):;
T2:=triangle(7/2,7/2-i*sqrt(2)/2,5/2+sqrt(2)/2-i*sqrt(2)/2):;T2;
T3:=triangle(7/2,7/2+i*sqrt(2)/2,5/2+sqrt(2)/2+i*sqrt(2)/2):;T3;
T4:=triangle(3/2,3/2+i*sqrt(2)/2,5/2-sqrt(2)/2+i*sqrt(2)/2):;T4;
T5:=triangle(3/2,3/2-i*sqrt(2)/2,5/2-sqrt(2)/2-i*sqrt(2)/2):;T5;
carre(3/2-i*sqrt(2)/2,3/2+a-i*sqrt(2)/2);
s:=segment(3/2+i*a-i*sqrt(2)/2,7/2-i*sqrt(2)/2):;s;
C1:=A1+i*sqrt(2)/2;
G1:=A1-i*sqrt(2)/2;
K1:=E1+i*sqrt(2)/2;
L1:=E1-i*sqrt(2)/2;
M1:=affichage(inter_unique(s,droite(y=sqrt(2)/2)),quadrant4);
N1:=inter_unique(s,droite(x=3/2+a));
```

```
O1:=inter_unique(s,droite(A1,H1));
Q1:=point(a+3/2-i*sqrt(2)/2));
C2:=translation(-affixe(C1)+a+3/2+i*a-i*sqrt(2)/2,evalf(C1));
C3:=translation(-a,evalf(C2));
```

On obtient :



On tape pour avoir le rectangle réalisé avec les 8 pièces en couleur :

```
P11:=polygone(D1,E1,F1,Q1,N1,M1):;
affichage(P11,1+rempli);
P12:=triangle(E1,D1,K1):;
affichage(P12,2+rempli);
P13:=triangle(E1,L1,F1):;
affichage(P13,3+rempli);
P14:=triangle(A1,B1,C1):;
affichage(P14,4+rempli);
P15:=triangle(A1,O1,G1):;
affichage(P15,5+rempli);
P16:=triangle(H1,O1,G1):;
affichage(P16,6+rempli);
P17:=polygone(Q1,H1,O1,N1):;
affichage(P17,47+rempli);
P18:=polygone(M1,O1,A1,B1):;
affichage(P18,148+rempli);
C2:=translation(-affixe(C1)+a+3/2+i*a-i*sqrt(2)/2,evalf(C1));
C3:=translation(-a,evalf(C2));
//rotation(evalf(G1),pi/2,evalf(P16));
```

On obtient les pièces du puzzle et le rectangle :



On tape :

```
carre(0,a):;
b:=3/2-i*sqrt(2)/2;
affichage(translation(-b,evalf(P11)),1+rempli);
affichage(translation(-b,evalf(P12)),2+rempli);
affichage(translation(-b,evalf(P13)),3+rempli);
affichage(translation(-b+-affixe(C1)+a+3/2+i*a-i*sqrt(2)/2,
                      evalf(P14)),4+rempli);
affichage(translation(-b+(N1-G1),evalf(P15)),5+rempli);
affichage(translation(-b+(C3-M1),evalf(P18)),148+rempli);
affichage(translation(-b+(M1-G1),evalf(P16)),6+rempli);
affichage(translation(-b+(M1-G1),evalf(P17)),47+rempli);
affichage(translation(1+i*sqrt(2)/2,evalf(P11)),1+rempli);
affichage(translation(1+i*sqrt(2)/2,evalf(P18)),148+rempli);
affichage(translation(1+i*sqrt(2)/2,evalf(P17)),47+rempli);
affichage(translation(7/2+i*sqrt(2)/2,evalf(P2)),2+rempli);
affichage(translation(7/2+i*sqrt(2)/2,evalf(P3)),3+rempli);
P4:=triangle(H,J,G):;
affichage(translation(7/2+i*sqrt(2)/2,evalf(P4)),4+rempli);
P5:=triangle(F,J,G):;
affichage(translation(7/2+i*sqrt(2)/2,evalf(P5)),5+rempli);
affichage(translation(3.5-affixe(G1),rotation(evalf(G1),
           pi/2,evalf(P16))),6+rempli);
```

On obtient l'octogone et le carré :



11.21 L'octogone et le carré

Voici le puzzle de Thomas Bennett (1868-1943) transformant un octogone en carré à l'aide de 5 pièces.



Si l'octogone s'inscrit dans un cercle de rayon 1 on a :
Soit l la longueur d'un côté de l'octogone et c la longueur d'un côté du carré.
On a $l = 2\sin(\pi/8 = \sqrt{2-\sqrt{2}}$ et $c^2 = 8 * (\sqrt{2}/4) = 2 * \sqrt{2}$ On tape :
`l:=sqrt(2-sqrt(2.)) c:=sqrt(2*sqrt(2.))`

11.21.1 Le découpage du carré

Pour découper le carré on a besoin des variables :
`a:=(c-l)/2`

`b:=(c+l)/2` et donc $b = a + l$
`d:=l/2`
`a1:=atan(c/l);`
`r1:=3*pi/8-a1`
On considère le carré de côté `c` et on reporte la longueur a sur chacun de ses côtés pour obtenir les points P, Q etc...(cf figure).
On découpe le carré de côté `c` à l'aide d'un carré `P1` de côté l et de 4 polygones identiques à `P2:=polygone(M,P,R,Q,N)` qui a comme côtés :
$MN = l$,$MP = NQ = d = l/2$,$PR = b$.
On remarquera que :
l'angle $a1 = \widehat{RPM} = \text{atan}(c/l)$ l'angle $\widehat{PMN} = \widehat{MNQ} = 3\pi/4$ On tape :

```
carre(0,c);
segment(a,b+c*i,affichage=2+ligne_tiret);
segment(c+a*i,b*i,affichage=2+ligne_tiret);
O:=point(c*(1+i)/2);
d:=longueur(a,O);
M:=point(a+(O-point(a))*l/d/2);
N:=point(c+a*i+(O-point(c+a*i))*l/d/2);
P:=point(a);
Q:=point(c+a*i);
R:=point(c);
P1:=carre(M,N);
P2:=polygone(M,P,R,Q,N);
P3:=rotation(O,pi/2,P2):;P3
P4:=rotation(O,pi,P2):;P4
P5:=rotation(O,-pi/2,P2):;P4
```

On obtient :



11.21.2 Les pièces en couleur du carré

On tape pour voir les pièces en couleur :

```
carre(0,c):;
segment(a,b+c*i);
segment(c+a*i,b*i);
O:=point(c*(1+i)/2);
d:=longueur(a,O);
M:=point(a+(O-point(a))*l/d/2):;
N:=point(c+a*i+(O-point(c+a*i))*l/d/2):;
P:=point(a):;
Q:=point(c+a*i):;
P1:=carre(M,N):;
affichage(P1,1+rempli);
P2:=polygone(M,P,point(c),Q,N):;
affichage(P2,2+rempli);
P3:=rotation(O,pi/2,P2):;
affichage(P3,3+rempli);
P4:=rotation(O,pi,P2):;
affichage(P4,4+rempli);
P5:=rotation(O,-pi/2,P2):;
affichage(P5,5+rempli);
```

On obtient :



Pour tracer l'octogone de centre O, on tape :

```
O:=point(c*(1+i)/2);
[A,B,C,D,E,F,G,H]:=(O+exp(i*k*pi/4))$(k=0..7);
polygone(A,B,C,D,E,F,G,H);
```

```
p1:=milieu(D,E);
translation(p1-P,P2)
r:=translation(p1-P,R)
```

On obtient :



On remarque qu'il faut faire subir à $P2$ une translation puis une rotation : en effet, le carré central n'a pas ses côtés parallèles à OA et OC car l'angle $\widehat{r, p1, D} = 3\pi/8 \neq a1$. On modifie donc :

```
O:=point(c*(1+i)/2);
[A,B,C,D,E,F,G,H]:=(O+exp(i*k*pi/4))$(k=0..7);
polygone(A,B,C,D,E,F,G,H);
p1:=milieu(D,E);
rotation(p1,3*pi/8-a1,translation(p1-P,P2))
```

On obtient :



On tape :

```
polygone(A,B,C,D,E,F,G,H):;
p1:=milieu(D,E):;
affichage(rotation(p1,r1,translation(p1-P,P2)),2+rempli);
p2:=milieu(B,C):;
affichage(rotation(p2,r1,translation(p2-point(b*i),P5)),5+rempli);
p3:=milieu(H,A):;
affichage(rotation(p3,r1,translation(p3-point(b+c*i),P4)),4+rempli);
p4:=milieu(F,G):;
affichage(rotation(p4,r1,translation(p4-point(c+a*i),P3)),3+rempli);
affichage(rotation(p4,r1,translation(p4-point(c+a*i),P3)),3+rempli);
K:=rotation(p3,r1,translation(p3-point(b+c*i),point(i*c)));
L:=rotation(p2,r1,translation(p2-point(b*i),point(0))):;
affichage(carre(K,L),1+rempli);
```

On obtient :



11.21.3 Le pavage induit par ce puzzle

On tape pour dessiner, à la position $z0$, le carré de côté c où l'on a accolé le carré de côté l en dessous de son coin inférieur droite :

```
bennett(z0):={
  local a,b,c,d,l,L,O,M,N,P,Q,M1,N1,P1,Q1;
  l:=sqrt(2-sqrt(2.));
  c:=sqrt(2*sqrt(2.));
  a:=(c-l)/2;
  b:=(c+l)/2;
  L:=NULL;
  L:=L,carre(z0,z0+c);
O:=point(z0+c*(1+i)/2);
d:=longueur(a+z0,O);
M:=point(z0+a+(O-point(z0+a))*l/d/2);
N:=point(z0+c+a*i+(O-point(z0+c+a*i))*l/d/2);
carre(M,N,M1,N1);
L:=L,carre(M,N);
P:=point(z0+a);
Q:=point(z0+c+a*i);
P1:=point(z0+(c-a)+c*i);
Q1:=point(z0+(c-a)*i);
L:=L,segment(M,P);
L:=L,segment(N,Q);
```

```
L:=L,segment(M1,P1);
L:=L,segment(N1,Q1);
L:=L,carre(z0+c,z0+c-l);
retourne L;
}:;
```

On tape :

```
bennett(0),bennett(c-i*l),
bennett(l+i*c),bennett(c+l+i*(c-l)),
bennett(2*c+l+i(c-2*l)),bennett(2*c-i*2*l),
bennett(c+2*l+i*(2*c-l)),bennett(2*c+2*l+i*(2*c-2l)),
bennett(3*c+l+i*(c-3*l)),bennett(3*c+2*l+i*(2*c-3*l))
```

On obtient :



Pour avoir le pavage en couleur, on tape :

```
bennettc(z0):={
  local a,b,c,d,l,L,O,M,N,P,Q,P2;
  l:=sqrt(2-sqrt(2.));
  c:=sqrt(2*sqrt(2.));
  a:=(c-l)/2;
  b:=(c+l)/2;
  L:=NULL;
O:=point(z0+c*(1+i)/2);
d:=longueur(a+z0,O);
M:=point(z0+a+(O-point(z0+a))*l/d/2);
N:=point(z0+c+a*i+(O-point(z0+c+a*i))*l/d/2);
L:=L,affichage(carre(M,N),1+rempli);
P:=point(z0+a);
Q:=point(z0+c+a*i);
P2:=polygone(M,P,point(z0+c),Q,N);
L:=L,affichage(P2,2+rempli);
L:=L,affichage(rotation(O,pi/2,P2),3+rempli);
L:=L,affichage(rotation(O,pi,P2),4+rempli);
L:=L,affichage(rotation(O,-pi/2,P2),5+rempli);
L:=L,affichage(carre(z0+c,z0+c-l),1+rempli);
retourne L;
}:;
```

On obtient :



11.22 Le puzzle : Henry Ernest Dudeney contre Sam Loyd

11.22.1 Le problème : équidécomposablité d'une mitre et d'un carré

Soit un carré $ABCD$ de centre O. On lui enlève le triangle AOD.
Trouver un puzzle, ayant un minimum de morceaux, qui transforme le polygone $ABCDO$ (qui ressemble à une mitre) en un carré.
Supposons que $AB = 2u$, le carré $ABCD$ a comme aire $4u^2$ et le polygone $ABCDO$ a comme aire $3u^2$. Donc le carré final a ses côtés de longueur $\sqrt{3}u$.
Sam Loyd a donné la solution suivante d'un puzzle de 4 pièces :



Dudeney lui a fait remarqué que son "carré" final n'était pas carré ! La figure obtenue par Loyd est en effet un rectangle de dimension :
7/4,12/7 c'est-à dire 1.75,1.71428571429 ce qui aprroche $\sqrt{3}$.
Dudeney a alors donné la solution suivante d'un puzzle de 5 pièces :



Pour l'instant, on n'a pas trouvé de découpage en 4 pièces.... Peut-être que vous allez un trouver un !
J'ai cherché mais je n'ai trouvé que d'autres découpages en 5 pièces.
En voici un voici :



11.22.2 Les commandes des figures

On tape dans un niveau de géométrie 2d pour avoir le résultat de Loyd :

```
polygone(0,2*i,1+i);
polygone(0,1/2,1/2+i/2,affichage=4+rempli);
```

```
polygone(2*i,1/2+2*i,1/2+3*i/2,affichage=5+rempli);
a:=1/4;b:=2/7;
L:=NULL;
pour k de 0 jusque 5 faire
  L:=L,2-k*a+i*(k+1)*b,2-(k+1)*a+i*(k+1)*b;
  L:=L,2-(k+1)*a+i*(k+1)*b,2-(k+1)*a+i*(k+2)*b;
fpour;
polygone(2,L,1/2+(2-b)*i,1/2+3/2*i,1+i,1/2+i/2,1/2,
         affichage=1+rempli);
polygone(L,1/2+2*i,2+2*i,affichage=2+rempli);
polygone(3+i,5/2+i,5/2+i/2,affichage=4+rempli);
polygone(3+i,5/2+i,5/2+3*i/2,affichage=5+rempli);
polygone(4,op(gauche([L],23)+[2$23]),5/2+(2-b)*i,5/2+3/2*i,
         3+i,5/2+i/2,5/2,affichage=1+rempli);
polygone(op([L1]+[(2+a-i*b)$24]),5/2+a+2*i-i*b,4+a+2*i-i*b,
         affichage=2+rempli);
```

On tape dans un niveau de géométrie 2d pour avoir le résultat de Dudeney :

```
segment(0,2*i);
polygone(0,1,1+i,affichage=4+rempli);
polygone(2*i,1+2*i,1+i,affichage=5+rempli);
polygone(1,2,2+i*sqrt(3),affichage=1+rempli);
G:=point(1+(3-sqrt(3))*i):;
segment(G,sqrt(3)+(3-sqrt(3))*i);
polygone(1,1+(3-sqrt(3))*i,sqrt(3)+(3-sqrt(3))*i,
         affichage=2+rempli);
polygone(2+sqrt(3)*i,2+2*i,1+2*i,1,1+(3-sqrt(3))*i,
         sqrt(3)+(3-sqrt(3))*i,affichage=3+rempli);
polygone(7/2+i*sqrt(3),5/2+i*sqrt(3),7/2+i*(sqrt(3)-1),
         affichage=5+rempli);
polygone(5/2+i*sqrt(3),5/2+i*(sqrt(3)-1),7/2+i*(sqrt(3)-1),
         affichage=4+rempli);
polygone(3/2+sqrt(3),5/2+sqrt(3),5/2+sqrt(3)+i*sqrt(3),
         affichage=1+rempli);
polygone(5/2+sqrt(3)+i*sqrt(3),7/2+sqrt(3)*i,
         7/2+(-3+2*sqrt(3))*i,affichage=2+rempli);
polygone(7/2+i*(sqrt(3)-1),5/2+(sqrt(3)-1)*i,5/2,3/2+sqrt(3),
         7/2+(-3+2*sqrt(3))*i,affichage=3+rempli);
```

On tape dans un niveau de géométrie 2d pour avoir mon découpage :

```
segment(0,2*i);
polygone(0,2,3/2+i*sqrt(3)/2,affichage=1+rempli);
polygone(5/2+i,5/2+sqrt(3),5/2,affichage=1+rempli);
polygone(0,3/2+i*sqrt(3)/2, (4+3*i-(sqrt(3)))/2,1/2+3/2*i,
         1+i,affichage=2+rempli);
C:=projection(droite(5/2+i,5/2+sqrt(3)),5/2+sqrt(3)+i*sqrt(3)):;
polygone(5/2+sqrt(3)+i*sqrt(3),5/2+i*sqrt(3),5/2+i,C,
         affichage=2+rempli);
```

```
polygone(2,2+i*3/2,2-sqrt(3)/2+i*3/2,affichage=3+rempli);
polygone(C,5/2+sqrt(3),5/2+sqrt(3)+i*sqrt(3),
         affichage=3+rempli);
polygone(2*i,1/2+3*i/2,2*i+1,affichage=4+rempli);
polygone(1+2*i,1/2+3*i/2,3/2*i+2,2+2*i,affichage=5+rempli);
F:=point(5/2+sqrt(3)-sqrt(2)/2+i*sqrt(3)):;
E:=rotation(5/2+sqrt(3)+i*sqrt(3),pi/12,F):;
D:=projection( droite(y=-(x-5/2)/sqrt(3)+1),E):;
polygone(5/2+sqrt(3)+i*sqrt(3),E,D,C,affichage=5+rempli);
triangle(E,D,rotation((D+E)/2,pi/2,E),affichage=4+rempli);
```

11.22.3 Puzzle transformant 1 rectangle en 1 rectangle de même aire

Dans la recherche du problème du puzzle Dudeney-Loyd, on peut facilement transformer le polygone $ABCOD$ en un rectangle de même aire, puis on peut transformer ce rectangle en un carré : c'est pourquoi on cherche comment faire un puzzle qui transforme 1 rectangle en un rectangle (ou un rectangle en un carré) de même aire.

Nous cherchons donc ici, comment on peut faire un puzzle qui réalise un rectangle de dimensions $a \times b$ (avec $b < a$) et un rectangle de dimensions $c \times d$ (avec $d \leq c < a$) et de même aire i.e. vérifiant $a * b = c * d$.
Voici 2 solutions de 3 pièces.
Première solution
On utilise la même méthode que Dudeney :
Soient $ABCD$ le rectangle de dimension $AB = a \times BC = b$ et $AB_1C_1D_1$ le rectangle de dimension $AB_1 = c \times B_1C_1 = d$.
Si $d <= 2b$ on peut faire la construction qui suit sinon si $2b < d <= 3b$, on transforme le rectangle $(a - c) \times b$ en le rectangle $c \times d_1$ avec $d1 = d - b$ ce qui oblige à avoir une pièce de plus qyi sera le rectangle $c \times b$.
Prenons comme repère $\overrightarrow{AB}, overrightarrowAD$.



B_1 est le point d'affixe c et D_1 est le point d'affixe $i * d$ dans ce repère.
La première pièce du puzzle est le triangle AB_1D (sur la figure en rouge).
Soit F le point d'affixe $a - c + i * b$ et G le point d'affixe $a - c + i * (2b - d)d$ La deuxième pièce du puzzle est le polygone B_1BCFG (sur la figure en vert).
La troisième pièce du puzzle est le triangle DFG (sur la figure en jaune).

On a donc $FG = b - (2b - d) = d - b$ et $DF = a - c = B_1B$, donc si on fait glisser la pièce verte le long de DB pour amener G en D (i.e on lui fait subir une translation de vecteur $\overrightarrow{GD}$) et glisser la pièce jaune le long de DB pour amener G en B_1 (i.e on lui fait subir une translation de vecteur $\overrightarrow{GB_1}$), on obtient un rectangle de dimension $c \times d$ (bien sûr si $c = \sqrt{a * b}$, on a alors $d = c$ et le rectangle de dimension $c \times d$ est un carré).
Voici le programme correspondant :

```
puzzler2r1(a,b,c):={
  local x0,d,p1,p2,p3,pc,pa,p11,p12,p13;
  si a<b alors (a,b):=(b,a);fsi;
  d:=a*b/c;
  si c<d alors (d,c):=(c,d);fsi;
  si a<c alors (a,c):=(c,a);(d,b):=(b,d);fsi;
  x0:=ceil(c);
  si d>2*b alors "utiliser puzzler2r1(a-c,b,c)" fsi;
  pc:=polygone(0,c,c+i*d,i*d):;
  pa:=polygone(x0,x0+a,x0+a+i*b,x0+i*b):;
  p1:=affichage(polygone(x0,x0+c,x0+i*b),1+rempli);
  p2:=affichage(polygone(x0+a-c+i*(2b-d),x0+c,x0+a,
                x0+a+i*b,x0+a-c+i*b),2+rempli);
  p3:=affichage(polygone(x0+i*b,x0+a-c+i*(2b-d),
                x0+a-c+i*b),3+rempli);
  p11:=affichage(polygone(0,0+c,i*b),1+rempli);
  p12:=affichage(polygone(i*b,2c-a+i*(d-b),c+i*(d-b),
                 c+i*d,i*d),2+rempli);
  p13:=affichage(polygone(2*c-a+i*(d-b),c,c+i*(d-b)),
                 3+rempli);
  retourne p1,p2,p3,p11,p12,p13;
  }:;
```

Deuxième solution
Soient $ABCD$ le rectangle de dimension $AB = a \times BC = b$ et $AB_1C_1D_1$ le rectangle de dimension $AB_1 = c \times B_1C_1 = d$. Si $\sqrt{a^2 - c^2} \leq d$ on peut faire la construction qui suit sinon si $d < \sqrt{a^2 - c^2} \leq 2d$, la même construction donnera 4 pièces et si $2d < \sqrt{a^2 - c^2} \leq 3d$, on transforme le rectangle $(a - c) \times b$ en le rectangle $c \times d_1$ avec $d1 = d - b$ ce qui oblige à avoir une pièce de plus qui sera le rectangle $c \times b$.
Prenons comme repère $\overrightarrow{AB}, \overrightarrow{AD}$



Soit E le point d'ordonnée positive qui est l'intersection du cercle de centre A et de rayon c avec le cercle de diamètre AD.
On aura 3 pièces si E se trouve à l'intérieur du rectangle $ABCD$ et sinon on aura 4 pièces.
E se trouve à l'intérieur du rectangle $ABCD$ si $\sqrt{a^2 - c^2} \leq d$ en effet on a :
aire de AEB= $a * y_E = c * \sqrt{a^2 - c^2}$ donc :
$y_E \leq b$ est équivalent à $ay_E = c * \sqrt{a^2 - c^2} \leq ab = cd$
ce qui est équivalent à $\sqrt{a^2 - c^2} \leq d$.
Si $\sqrt{a^2 - c^2} \leq d$: on obtient 3 pièces.
La première pièce du puzzle est le triangle rectangle AED (sur la figure en rouge).
La deuxième pièce du puzzle est le polygone $AEFD$ (sur la figure en vert).
La troisième pièce du puzzle est le triangle rectangle ACF (sur la figure en jaune) dans lequel $FB = AD = d$ $HG = DF$.
Sur la figure de droite on voit comment les 2 rectangles sont imbriqués.
Si $2d \geq \sqrt{a^2 - c^2} > d$: on obtient 4 pièces.
Si $3d \geq \sqrt{a^2 - c^2} > 2d$ on Voici la figure des 2 rectangles imbriqués :



Voici les programmes qui correspondent aux 2 cas $\sqrt{a^2-c^2} \leq d$ et $\sqrt{a^2-c^2} > d$:

```
puzzler2r2(a,b,c):={
  local x0,d,p1,p2,p3,pc,pa,p11,p12,p13,l;
  si a<b alors (a,b):=(b,a);fsi;
  d:=a*b/c;
  si c<d alors (d,c):=(c,d);fsi;
  si a<c alors (a,c):=(c,a);(d,b):=(b,d);fsi;
  si sqrt(a^2-c^2)>d alors
     retourne puzzler2r21(a,b,c);
  fsi;
  x0:=ceil(c);
  l:=sqrt(a^2-c^2);
  pc:=polygone(0,c,c+i*d,i*d);
  pa:=polygone(x0,x0+a,x0+a+i*b,x0+i*b);
  p1:=affichage(polygone(x0,x0+a,x0+c^2/a+i*c*l/a),
                 1+rempli);
  p2:=affichage(polygone(x0,x0+i*b,x0+a-b*l/c+i*b,
                 x0+c^2/a+i*c*l/a),2+rempli);
  p3:=affichage(polygone(x0+a,x0+a+i*b,(x0+a-b*l/c)+i*b),
                 3+rempli);
  p11:=affichage(polygone(i*d,c+i*(d-l),c+i*d),1+rempli);
  p12:=affichage(polygone(0,c,c+i*(d-l),b*l/a+i*c*b/a),
                  2+rempli);
  p13:=affichage(polygone(0,i*d,b*l/a+i*c*b/a),3+rempli);
  retourne p1,p2,p3,p11,p12,p13;
  }:;
puzzler2r21(a,b,c):={
  local x0,d,p1,p2,p3,p4,pc,pa,p11,p12,p13,p14,l,E,M,F,P,Q;
```

```
  si a<b alors (a,b):=(b,a);fsi;
  d:=a*b/c;
  si c<d alors (d,c):=(c,d);fsi;
  si a<c alors (a,c):=(c,a);(d,b):=(b,d);fsi;
  si sqrt(a^2-c^2)<=d alors retourne puzzler2r2(a,b,c); fsi;
  x0:=ceil(c);
  l:=sqrt(a^2-c^2);
  E:=point(c^2/a+i*c*l/a);
  F:=point(a-b*l/c +i*b);
  M:=point(b*c/l+i*b);
  Q:=a+(E-F);
  P:=a+(M-F);
  pc:=polygone(0,c,c+i*d,i*d);
  pa:=polygone(x0,x0+a,x0+a+i*b,x0+i*b);
  p1:=affichage(polygone(x0,x0+affixe(M),x0+affixe(F),
              x0+affixe(Q),x0+affixe(P)),1+rempli);
  p2:=affichage(polygone(x0,x0+i*b,x0+affixe(M)),
              2+rempli);
  p3:=affichage(polygone(x0+a,x0+affixe(F),
              x0+a+i*b),3+rempli);
  p4:=affichage(polygone(x0+a,x0+affixe(Q),
              x0+affixe(P)),4+rempli);
  p11:=affichage(polygone(i*d+longueur(0,M),c+i*d-i*longueur(E,F),
               c,c-longueur(P,Q)),i*d,1+rempli);
  p12:=affichage(polygone(0,b*l/a+i*b*c/a,c-longueur(P,Q)),
               2+rempli);
  p13:=affichage(polygone(0,i*d,b*l/a+i*b*c/a,3+rempli);
  p14:=affichage(polygone(c+i*d,longueur(0,M)+i*d,
               c+i*d-i*longueur(E,F)),4+rempli);
retourne p1,p2,p3,p4,p11,p12,p13,p14;
  }:;
```

11.22.4 D’autres puzzles de 5 pièces

On va transformer de différentes façons le polygone $ABCOD$ en un rectangle d’aire égale à 3, puis on transfermera ce rectangle en un carré de côté $\sqrt{3}$.

Les différentes transformations de $ABCOD$ en un rectangle $A_1B_1C_1D_1$

Voici 4 transformations possibles (bien sûr il y en d’autres !) :



Pour le quatrième, il faut admettre que les pièces sont réversibles.
On cherche maintenant des puzzles de 5 pièces.

Avec le premier découpage

C'est celui de Dudeney : il a ensuite transformer le rectangle en un carré avec la 1ière méthode (la 2ième méthode est possible mais elle va donner un puzzle avec 6 pièces).

Avec le deuxième découpage

Avec la première méthode, on obtient :



Avec la deuxième méthode, on obtient :



Avec le troisième découpage
Avec la première méthode, on obtient :



Avec la deuxième méthode, on obtient le puzzle donné au debut de la section.
Avec le quatrième découpage
On uitilise la première méthode et à condition que la pièce verte soit réversible, on obtient :



si la pièce verte n'est pas réversible, il faut rajouté une sixième pièce :



On a tapé pour ces 2 dernières figures :

```
segment(3-sqrt(3)+i*(-sqrt(3)+2), 3-sqrt(3)+i);
polygone(0,sqrt(3),i,affichage=1+rempli);
polygone(7/2,7/2+sqrt(3),7/2+i,affichage=1+rempli);
polygone(2,3,3+i,3-sqrt(3)+i,3-sqrt(3)+i*(sqrt(3)-1),
  affichage=2+rempli);
polygone(3-sqrt(3)+i*(2-sqrt(3)),sqrt(3),2,
  3-sqrt(3)+i*(sqrt(3)-1),affichage=5+rempli);
t1:=1/2+sqrt(3)+i*(sqrt(3)-1);
polygone(t1+3-sqrt(3)+i*(2-sqrt(3)),t1+sqrt(3),t1+2,
  t1+3-sqrt(3)+i*(sqrt(3)-1),affichage=5+rempli);
polygone(t1+2,t1+3,t1+3+i,t1+3-sqrt(3)+i,
  t1+3-sqrt(3)+i*(sqrt(3)-1),affichage=2+rempli);
polygone(3-sqrt(3)+i,3-sqrt(3)+i*(sqrt(3)-1),1+i,
  affichage=4+rempli);
polygone(3-sqrt(3)+i*(2-sqrt(3)),3-sqrt(3)+i*(sqrt(3)-1),
  1+i,i,affichage=3+rempli);
t2:=1/2+2*sqrt(3)+i*(sqrt(3)-2);
polygone(t2+3-sqrt(3)+i*(2-sqrt(3)),t2+3-sqrt(3)+i*(sqrt(3)-1),
  t2+1+i,t2+i,affichage=3+rempli);
polygone(t2+3-sqrt(3)+i,t2+3-sqrt(3)+i*(sqrt(3)-1),
  t2+1+i,affichage=4+rempli);
polygone(-9/4*i+2,-9/4*i+2+sqrt(3),i-9/4*i+2,
  affichage=1+rempli);
t3:=-9/4*i+2;
```

```
polygone(-9/4*i+5-sqrt(3)+i*(2-sqrt(3)),-9/4*i+5-sqrt(3)+
  i*(sqrt(3)-1),-9/4*i+3+i,-9/4*i+2+i,affichage=3+rempli);
polygone(t3+3-sqrt(3)+i*(2-sqrt(3)),t3+sqrt(3),t3+2,
  t3+3-sqrt(3)+i*(sqrt(3)-1),affichage=5+rempli);
polygone(-5/4*i+3,-5/4*i+2,-1/4*i+2,-1/4*i+2+sqrt(3),
  2+sqrt(3)-1/4*i+i*(sqrt(3)-2),affichage=2+rempli);
polygone(-1/4*i+4,-1/4*i+2+sqrt(3),2+sqrt(3)-1/4*i+i*(sqrt(3)-2),
  affichage=4+rempli);
```

que l'on a modifié pour mettre la 6ième pièce :

```
segment(3-sqrt(3)+i*(-sqrt(3)+2), 3-sqrt(3)+i);
polygone(0,sqrt(3),i,affichage=1+rempli);
polygone(7/2,7/2+sqrt(3),7/2+i,affichage=1+rempli);
polygone(2,3+i,3-sqrt(3)+i,3-sqrt(3)+i*(sqrt(3)-1),
  affichage=2+rempli);
polygone(2,3,3+i,affichage=6+rempli);
polygone(3-sqrt(3)+i*(2-sqrt(3)),sqrt(3),2,
  3-sqrt(3)+i*(sqrt(3)-1),affichage=5+rempli);
t1:=1/2+sqrt(3)+i*(sqrt(3)-1);
polygone(t1+3-sqrt(3)+i*(2-sqrt(3)),t1+sqrt(3),
  t1+2,t1+3-sqrt(3)+i*(sqrt(3)-1),affichage=5+rempli);
polygone(t1+2,t1+3+i,t1+3-sqrt(3)+i,t1+3-sqrt(3)+i*(sqrt(3)-1),
  affichage=2+rempli);
polygone(t1+2,t1+3,t1+3+i,affichage=6+rempli);
polygone(3-sqrt(3)+i,3-sqrt(3)+i*(sqrt(3)-1),1+i,
  affichage=4+rempli);
polygone(3-sqrt(3)+i*(2-sqrt(3)),3-sqrt(3)+i*(sqrt(3)-1),
  1+i,i,affichage=3+rempli);
t2:=1/2+2*sqrt(3)+i*(sqrt(3)-2);
polygone(t2+3-sqrt(3)+i*(2-sqrt(3)),t2+3-sqrt(3)+i*(sqrt(3)-1),
  t2+1+i,t2+i,affichage=3+rempli);
polygone(t2+3-sqrt(3)+i,t2+3-sqrt(3)+i*(sqrt(3)-1),t2+1+i,
  affichage=4+rempli);
polygone(-9/4*i+2,-9/4*i+2+sqrt(3),i-9/4*i+2,affichage=1+rempli);
t3:=-9/4*i+2;
polygone(-9/4*i+2+3-sqrt(3)+i*(2-sqrt(3)),-9/4*i+2+3-sqrt(3)+i*(sqrt(3)-1),
  -9/4*i+2+1+i,-9/4*i+2+i,affichage=3+rempli);
polygone(t3+3-sqrt(3)+i*(2-sqrt(3)),t3+sqrt(3),t3+2,
  t3+3-sqrt(3)+i*(sqrt(3)-1),affichage=5+rempli);
t0:=1-i*5/4;
polygone(t0+2,t0+3+i,t0+3-sqrt(3)+i,t0+3-sqrt(3)+i*(sqrt(3)-1),
  affichage=2+rempli);
polygone(t0+1,t0+2,t0+1+i,affichage=6+rempli);
polygone(t0+3-sqrt(3)+i,t0+3-sqrt(3)+i*(sqrt(3)-1),t0+1+i,
  affichage=4+rempli);
```

11.22.5 Une solution de 6 pièces et de 8 pièces 2 à 2 égales

On utilise la relation :
$\sqrt{3}$ est la longueur de l'hypoténuse d'un triangle rectangle de côté 1 et $\sqrt{2}$ (car $(\sqrt{3})^2 + 3 = 1 + 2 = 1^2 + (\sqrt{2})^2$: attention ce triangle n'est pas la moitié d'un triangle équilatéral !) L'aire de 4 de ces triangles rectangles vaut $2\sqrt{2}$ il manque alors un carré d'aire $3 - 2\sqrt{2}$ i.e . de côté $\sqrt{2} - 1$ car :
$(\sqrt{2} - 1)^2 = 3 - 2\sqrt{2}$
On utilise pas un rectangle comme inremédiaire mais 2 pièces doivent avoir une double face i.e réversible.
On obtient un puzzle de 6 pièces dont 2 pièces la verte et la bleue sont à double face i.e réversible :



On a tapé :

```
polygone(0,1,sqrt(2)*i,affichage=1+rempli);
polygone(2,1,2+sqrt(2)*i,affichage=2+rempli);
polygone(1,1+i,2-sqrt(2)+sqrt(2)*i,sqrt(2)*i,affichage=3+rempli);
polygone(1,1+i,sqrt(2)+sqrt(2)*i,2+sqrt(2)*i,affichage=4+rempli);
polygone(2*i,sqrt(2)*i,2-sqrt(2)+sqrt(2)*i,affichage=5+rempli);
polygone(2+2*i,2+sqrt(2)*i,sqrt(2)+sqrt(2)*i,affichage=6+rempli);
A:=point(5/2+sqrt(3)/3+i*sqrt(2/3)):;
polygone(5/2,5/2+sqrt(3),A,affichage=1+rempli);
B:=point(5/2+(3-sqrt(2))/sqrt(3)+i/sqrt(3)):;
carre(A,B,C,D);
F:=rotation(A,pi/2,C):; G:=rotation(C,pi/2,A):;
polygone(5/2+i*sqrt(3),C,5/2+sqrt(3)+i*sqrt(3),affichage=2+rempli);
polygone(5/2+i*sqrt(3),5/2,A,F,affichage=3+rempli);
polygone(5/2+sqrt(3),G,B,5/2+sqrt(3)+i*sqrt(3),affichage=4+rempli);
polygone(A,C,F,affichage=5+rempli);
polygone(A,C,G,affichage=6+rempli);
```

Un autre puzzle de 8 pièces égales 2 à 2.
On utilise le triangle rectangle de côtés 1, $\sqrt{2}$, $sqrt3$ (attention ce n'est pas la moitié d'un triangle équilatéral !) :



On a tapé pour faire ce dessin :

```
polygone(0,1+i,1-1/sqrt(2)+i*(1+1/sqrt(2)),affichage=1+rempli);
polygone(2+2*i,1+i,(1+1/sqrt(2))+i*(1-1/sqrt(2)),affichage=2+rempli);
polygone(0,1+i,1,affichage=3+rempli);
polygone(2,1+i,1,affichage=4+rempli);
polygone(0,sqrt(2)*i,1-1/sqrt(2)+i*(1+1/sqrt(2)),affichage=5+rempli);
polygone(2+2*i,(1+1/sqrt(2))+i*(1-1/sqrt(2)),2+(2-sqrt(2))*i,affichage=6+remp
polygone(2*i,sqrt(2)*i,1-1/sqrt(2)+i*(1+1/sqrt(2)),affichage=0+rempli);
polygone(2,2+(2-sqrt(2))*i,1+1/sqrt(2)+i*(1-1/sqrt(2)),affichage=17+rempli);
B:=rotation(5/2+sqrt(3)*(1+i),atan(sqrt(2)),5/2+sqrt(3)-1+sqrt(3)*i):;
C:=rotation(5/2+sqrt(3),atan(sqrt(2)),5/2+sqrt(3)+i):;
polygone(5/2+sqrt(3),5/2+sqrt(3)+sqrt(3)*i, C,affichage=1+rempli);
polygone(5/2+i*sqrt(3),5/2+sqrt(3)*(1+i), B,affichage=2+rempli);
P:=projection(droite(5/2+sqrt(3)*i,B),5/2):;
polygone(P,5/2+sqrt(3)*i,rotation(P,pi/2,(sqrt(3)*i+5/2)),affichage=3+rempli)
Q:=projection(droite(5/2+sqrt(3),C),P):;
polygone(Q,5/2,rotation(Q,pi/2,5/2),affichage=4+rempli);
R:=inter_unique(cercle(5/2,sqrt(2)-1),cercle(5/2+i*sqrt(3),sqrt(2))):;
polygone(5/2,5/2+i*sqrt(3), R,affichage=6+rempli);
S:=inter_unique(cercle(5/2,sqrt(2)),cercle(5/2+sqrt(3),sqrt(2)-1)):;
polygone(5/2,5/2+sqrt(3), S,affichage=5+rempli);
polygone(B,Q,C,affichage=17+rempli);
polygone(B,P,Q,affichage=0+rempli);
```

11.22.6 Une autre solution de 6 pièces sans un rectangle comme intermédiaire



On a tapé pour faire ce dessin :

```
polygone(i*(2-sqrt(3)),2*i,1+i,1+i*(2-sqrt(3)),affichage=1+rempli);
polygone(5/2+i*(2-sqrt(3)),5/2+2*i,5/2+1+i,5/2+1+i*(2-sqrt(3)),afficha
segment((2-sqrt(3))*i,2+(2-sqrt(3))*i);
polygone(5/2-i,5/2-i+2,5/2-i+2+i*(2-sqrt(3)),5/2-i+(2-sqrt(3))*i);
polygone(1+i*(2-sqrt(3)),2+i*(2-sqrt(3)),2+2*i,1+i,1+i*(2-sqrt(3)),af
polygone(5/2+sqrt(3)+2*i,5/2+2*i,5/2+1+i,5/2+sqrt(3)+i,affichage=2+rem
carre(-i,sqrt(3)-1-i);
segment(sqrt(3)-1-i+(2*sqrt(3)-3)*i, -i+(2*sqrt(3)-3)*i);
segment(-i+5/2+3-sqrt(3),-i+5/2+3-sqrt(3)+i*(2-sqrt(3)));
s:=segment(-i+5/2+i*(2-sqrt(3)),-i+5/2+sqrt(3)-1);
A:=inter_unique(droite(x=5/2+4-2*sqrt(3)),s):;
B:=inter_unique(droite(x=5/2+4-2*sqrt(3)),droite(y=-1+2-sqrt(3))):;
segment(A,B);
segment(-i+i*(2-sqrt(3)),-i+sqrt(3)-1);
segment(2*c-a+i*(d-b)-i,c+i*(d-b)-i);
segment(-i+5/2+a-c+i*(2*b-d),-i+5/2+i*b);
polygone(2,2+i*(2-sqrt(3)),3-sqrt(3)+ i*(2-sqrt(3)),3-sqrt(3),
         affichage=3+rempli);
polygone(5/2+sqrt(3)+i*(-1+sqrt(3)),5/2+sqrt(3)+i,5/2+1+i,
         5/2+1+i*(-1+sqrt(3)),affichage=3+rempli);
polygone(0,-1+sqrt(3),i*(2-sqrt(3)),affichage=4+rempli);
polygone(7/2+i*(2-sqrt(3)),7/2-1+sqrt(3)+i*(2-sqrt(3)),
         7/2+2*i*(2-sqrt(3)),affichage=4+rempli);
polygone(4-2*sqrt(3)+i*(7-4*sqrt(3)),4-2*sqrt(3)+i*(2-sqrt(3)),
         i*(2-sqrt(3)),affichage=5+rempli);
polygone(5/2+sqrt(3)+i*(2-sqrt(3)),5/2+sqrt(3)+i*(-3+2*sqrt(3)),
```

```
          5/2-4+3*sqrt(3)+i*(-3+2*sqrt(3)),affichage=5+rempli);
polygone(4-2*sqrt(3)+i*(2-sqrt(3)),4-2*sqrt(3)+i*(7-4*sqrt(3)),
          sqrt(3)-1,3-sqrt(3),3-sqrt(3)+i*(2-sqrt(3)),
          affichage=6+rempli);
polygone(5/2+1+i*(4-2*sqrt(3)),5/2+1+i*(-1+sqrt(3)),
          5/2+sqrt(3)+i*(-1+sqrt(3)),5/2+sqrt(3)+i*(-3+2*sqrt(3)),
          5/2-4+3*sqrt(3)+i*(-3+2*sqrt(3)),affichage=6+rempli);
```

11.23 Puzzle d'un triangle quelconque et d'un carré et d'un rectangle

On peut faire un puzzle qui reconstitue soit un triangle quelconque de surface S, soit un rectangle de dimension $a\ x\ b = S/a$ et bien sur ce rectangle sera un carré de côté l si $a = \sqrt{S}$.
On va faire ici la construction du puzzle pour transformer un triangle quelconque en un carré.

11.23.1 La construction du puzzle

Soit un triangle ABC : on choisit $AB = 10$ et C d'affixe $a + ib$ (la figure ci est faite avec $a = 2$ et $c = 7$).
Soient I (resp J, K) le milieu de AB (resp BC, AC).
ABC a comme aire $S = 5b$ donc le carré aura comme côté $l = \sqrt{5b}$ Soit P le point du segment BC tel que $IP = l$ et Q le point du segment BC tel que $PQ = KI = BC/2$. Le quadrilatère $PKIQ$ est un parallélogramme. Les triangles PKJ et QIB sont égaux.
Les triangles CKJ et KAI sont égaux et ils ont comme aire $S/4$ Donc l'aire du parallélogramme $PKIQ$ vaut $S/2$.
Soient L (resp H) la projection de K (resp Q) sur PI. Les triangles PKL et IQH sont égaux, donc les aires des triangles PKI et IQP valent $S/4$ ce qui implique que $KL = HQ = l/2$
Soit D le symétrique de C par rapport à P. Puisque $PQ = BC/2$ on a $DQ = QB$ et donc D est le symétrique de B par rapport à Q.
Attention On suppose que P (tel que $IP = l$) est sur CJ et que L la projection de K sur PI est sur PI : c'est le cas si les 3 angles du triangle ABC sont aigus...



Le puzzle a 4 pièces qui sont :

```
p1:=polygone(A,I,L,K,affichage=1+rempli);
p2:=polygone(I,B,Q,H,affichage=2+rempli);
p3:=polygone(P,C,K,L,affichage=3+rempli);
p4:=triangle(P,H,Q,affichage=4+rempli);
```

On obtient le carré faisant effectuer :
à $p1$ une translation de vecteur $\overrightarrow{AD}$ ($\overrightarrow{DM} = \overrightarrow{AK} = \overrightarrow{KC}$ et, $\overrightarrow{DN} = \overrightarrow{AI} = \overrightarrow{IB}$)
à $p2$ une symétrie par rapport à Q et
à $p3$ une symétrie par rapport à P.
On tape pour obtenir la figure ci-dessus :

```
A:=point(0);;
B:=point(10);
supposons(a=[2,-5,5,0.1]);
supposons(b=[7,-5,5,0.1]);
C:=point(a+i*b);
l:=sqrt(5*b);
I:=point(5);
J:=milieu(B,C);
K:=milieu(A,C);
triangle(A,B,C);
s:=segment(B,C):;
P:=inter_unique(cercle(I,l),s);
segment(K,P,affichage=3);segment(I,P);
L:=projection(segment(I,P),K);
segment(K,L);
d:=parallele(I,segment(K,P)):;
```

```
Q:=inter_unique(d,s);
H:=projection(segment(I,P),Q);
polygone(Q,P,K,I);
segment(Q,H);
D:=symetrie(P,C);
translation(D-A,p1);
symetrie(Q,p2);
symetrie(P,p3);
S:=translation(D-A,L);
R:=symetrie(Q,H);
T:=symetrie(P,L);
M:=symetrie(P,K);
N:=symetrie(Q,I);
```

11.23.2 Le pavage induit



On tape :

```
supposons(a=[2.0,-5,5,0.1]);
supposons(b=[7,-5,5,0.1]);
p1:=polygone(A,I,L,K,affichage=1+rempli);
p2:=polygone(I,B,Q,H,affichage=2+rempli);
p3:=polygone(P,C,K,L,affichage=3+rempli);
p4:=triangle(P,H,Q,affichage=4+rempli);
c3:=affichage(symetrie(P,p3),3+rempli));
c2:=affichage(symetrie(Q,p2),2+rempli));
c1:=affichage(translation(D-A,p1),1+rempli));
affichage(translation(D-A,p3),3+rempli));
```

```
affichage(translation(D-A,p2),2+rempli));
affichage(translation(D-A,p4),4+rempli));
affichage(translation(D-A,c1),1+rempli);
affichage(translation(D-A,c2),2+rempli);
affichage(translation(D-A,c3),3+rempli);
sI:=symetrie(P,I):;;
affichage(symetrie(N,c1),1+rempli));
s4:=affichage(symetrie(Q,p4),4+rempli));
T:=symetrie(P,L):;S:=translation(D-A,L):;
sS:=symetrie(N,S):;
affichage(translation(sS-T,c3),3+rempli);
affichage(symetrie(P,p4),4+rempli));
affichage(symetrie(M,c1),1+rempli));
affichage(translation(sI-N,c2),2+rempli);
affichage(translation(D-A,s4),4+rempli);
affichage(symetrie(N,c3),3+rempli);
```

On obtient la figure ci-dessus.

11.23.3 Le cas du triangle équilatéral

On retrouve le puzzle de Dudeney et le pavage induit si le triangle ABC est équilatéral i.e. si $a = 5$ et $b = 5 * sqrt(3) \simeq 8.$, on obtient :



11.23.4 Le cas du triangle isocèle rectangle

Si le triangle ABC est isocèle rectangle :
si $a = 5$ et $b = 5$, on obtient :



si $a = 0$ et $b = 10$, on obtient :



11.23.5 Avec une animation

On tape (il faut avoir dédfini `p1,p2,p3,p4`) comme ci dessus) :

```
animtri(t1,t2,t3):={
```

```
local L;
L:=NULL;
L:=L,affichage(p4,4+rempli);
L:=L,affichage(rotation(Q,t2*pi*1.,p2),2+rempli);
L:=L,affichage(rotation(P,-t3*pi*1.,p3),3+rempli);
L:=L,affichage(translation(t1*(D-A),p1),1+rempli);
return L;
}:;

T2:=seq([animtri(0,t2,0)],t2=0.0..1.0,0.1):;
T3:=seq([animtri(0,1,t3)],t3=0.0..1.0,0.1):;
T1:=seq([animtri(t1,1,1)],t1=0.0..1.0,0.1):;
T5:=seq([animtri(1,t2,1)],t2=1.0..0.0,0.1):;
T6:=seq([animtri(1,0,t3)],t3=1.0..0.0,0.1):;
T4:=seq([animtri(t1,0,0)],t1=1.0..0.0,0.1):;
```

Puis dans un niveau de géométrie, on tape :
`animation(T2,T3,T1,T5,T6,T4)`
On lance l'animation avec le bouton `M` situé dans le pavé à droite de lécran graphique, puis on sélectionne `Animation->Créer animation`.
On arrête avec `Animation->Fin`

11.23.6 Avec une autre animation

On tape (ici tout à été redéfini) :

```
animtrig(a):={
local A,B,C,D,I,K,L,H,P,Q,l,LR,d,s,p4,I1,A1,L1,K1,P1,C1,C2,P2,L2,K2;
LR:=NULL;
C:=point(2+i*7);
A:=point(0);
B:=point(10);
I:=point(5);
K:=point(1+i*3.5);
l:=sqrt(35);
s:=segment(B,C);
P:=evalf(inter_unique(cercle(I,l),s));
L:=evalf(projection(segment(I,P),K));
d:=parallele(I,segment(K,P));
Q:=evalf(inter_unique(d,s));
H:=evalf(projection(segment(I,P),Q));
D:=evalf(symetrie(P,C));
p4:=affichage(triangle(P,H,Q),4+rempli);
LR:=LR,p4;
LR:=LR,affichage(rotation(Q,a,quadrilatere(I,B,Q,H)),2+rempli);
I1:=rotation(Q,a,I);
A1:=rotation(Q,a,A);
L1:=rotation(Q,a,L);
K1:=rotation(Q,a,K);
```

```
C1:=rotation(Q,a,C);
P1:=rotation(Q,a,P);
LR:=LR,affichage(rotation(I1,a,evalf(quadrilatere(A1,I1,L1,K1))),1+rempli);
P2:=rotation(I1,a,P1);
C2:=rotation(I1,a,C1);
L2:=rotation(I1,a,L1);
K2:=rotation(I1,a,K1)
LR:=LR,affichage(rotation(K2,a,evalf(quadrilatere(P2,C2,K2,L2))),3+rempli);
return LR;
 }:;
T1:=seq([animtrig(t4)],t4=0..3.14,3.14/10):;
T2:=seq([animtrig(t4)],t4=3.14..0,3.14/10):;
```

Puis dans un niveau de gépmétrie, on tape :
`animation(T1,T2)`
On obtient :



11.24 Puzzle d'un demi hexagone d'un triangle et d'un demi carré

On transforme un demi hexagone de côté r en le triangle ABC $AB = 3$, $AC = r$ et $\widehat{BAC} = \pi/3$ de même aire $S = 3r^2\sqrt{3}/4$).
Puis on transforme le triangle ABC en un demi carré de dimention $l\,x/2$ ($KP = l$ avec `l:=evalf(sqrt(3*sqrt(3)/2));`) selon la méthode qui transforme un triangle quelconque en un rectangle (11.23).
On tape :

```
A:=point(0);
B:=point(3);
C:=point(1/2,sqrt(3)/2);
```

```
hexagone(C,A);
s:=segment(B,A);segment(B,C);
I:=point(3/2);
J:=milieu(B,C));
K:=milieu(C,A));
l:=evalf(sqrt(3*sqrt(3)/2));
P:=inter_unique(cercle(K,l),s);
segment(P,K);
M:=projection(droite(K,P),J);
Q:=inter_unique(parallele(K,droite(J,P)),s);
N:=projection(droite(K,P),Q);
H:=point(2);
p1:=polygone(J,M,P,H):;p1;
p2:=polygone(K,A,Q,N):;p2;
p3:=polygone(K,M,J,C):;p3;
H2:=point(3/2,sqrt(3)/2):;;
p4:=triangle(J,H,B):;p4;
p5:=triangle(P,N,Q):;p5;
D:=symetrie(P,B);
L:=milieu(K,M);
```

On obtient :



On tape pour faire un demi carré et un demi hexagone avec 5 pièces :

```
affichage(p1,1+rempli);
affichage(p2,2+rempli);
affichage(p3,3+rempli);
affichage(p4,4+rempli);
affichage(p5,5+rempli);
c3:=translation(D-C,p3):;;
```

```
c1:=symetrie(P,p1):;;
c4:=symetrie(P,p4):;;
c2:=symetrie(Q,p2):;;
affichage(translation(-i,p5),5+rempli);
affichage(translation(-i,c1),1+rempli);
affichage(translation(-i,c2),2+rempli);
affichage(translation(-i,c3),3+rempli);
affichage(translation(-i,c4),4+rempli);
R:=point(3-2*i+sqrt(3)*i):;;
affichage(translation(R-C,p3),3+rempli);
affichage(translation(R-C,p1),1+rempli);
affichage(translation(R-C,symetrie(J,p4)),4+rempli);
affichage(translation(R-C,p2),2+rempli);
affichage(translation(R-C,p5),5+rempli);
```

On obtient :



On tape pour faire un carré et un hexagone avec 2 fois 5 pièces :

```
affichage(p1,1+rempli);
affichage(p2,2+rempli);
affichage(p3,3+rempli);
affichage(p4,4+rempli);
affichage(p5,5+rempli);
c3:=translation(D-C,p3):;
c1:=symetrie(P,p1):;
c4:=symetrie(P,p4):;
c2:=symetrie(Q,p2):;
affichage(translation(-i,p5),5+rempli);
affichage(translation(-i,c1),1+rempli);
affichage(translation(-i,c2),2+rempli);
affichage(translation(-i,c3),3+rempli);
```

```
affichage(translation(-i,c4),4+rempli);
R:=point(3-2*i+sqrt(3)*i):;;
affichage(translation(R-C,p3),3+rempli);;
affichage(translation(R-C,p1),1+rempli);;
affichage(translation(R-C,symetrie(J,p4)),4+rempli);
affichage(translation(R-C,p2),2+rempli);
affichage(translation(R-C,p5),5+rempli);
O:=point(7/2-2*i+sqrt(3)/2*i):;;
affichage(symetrie(O,translation(R-C,p5)),5+rempli);
affichage(symetrie(O,translation(R-C,p1)),1+rempli);
affichage(symetrie(O,translation(R-C,p2)),2+rempli);
affichage(symetrie(O,translation(R-C,p3)),3+rempli);
affichage(symetrie(O,translation(R-C,symetrie(J,p4))),4+rempli);
T:=translation(-i,symetrie(Q,N)):;
S:=translation((D-C)-i,M):;
affichage(symetrie(milieu(T,S),translation(-i,c2)),2+rempli);
affichage(symetrie(milieu(T,S),translation(-i,c1)),1+rempli);
affichage(symetrie(milieu(T,S),translation(-i,c3)),3+rempli);
affichage(symetrie(milieu(T,S),translation(-i,p5)),5+rempli);
affichage(symetrie(milieu(T,S),translation(D-C-i,symetrie(J,p4))),4+re
```

On obtient :



ou bien on remplace les 6 dernières lignes :

```
S:=translation((D-C)-i,M):;
affichage(symetrie(milieu(T,S),translation(-i,c2)),2+rempli);
affichage(symetrie(milieu(T,S),translation(-i,c1)),1+rempli);
affichage(symetrie(milieu(T,S),translation(-i,c3)),3+rempli);
affichage(symetrie(milieu(T,S),translation(-i,p5)),5+rempli);
affichage(symetrie(milieu(T,S),translation(D-C-i,symetrie(J,p4))),4+re
```

par :

```
U:=translation(-i,N):;
affichage(translation((T-U)-i,c2),2+rempli);
affichage(translation((T-U)-i,c1),1+rempli);
affichage(translation((T-U)-i,c3),3+rempli);
affichage(translation((T-U)-i,p5),5+rempli);
affichage(translation((T-U)+(D-C)-i,symetrie(J,p4)),4+rempli);
```

On obtient :



On tape pour 10 pièces (5+5 symétriques)

```
affichage(p1,1+rempli);
affichage(p2,2+rempli);
affichage(p3,3+rempli);
affichage(p4,4+rempli);
affichage(p5,5+rempli);
c3:=translation(D-C,p3):;
c1:=symetrie(P,p1):;
c4:=symetrie(P,p4):;
c2:=symetrie(Q,p2):;
affichage(translation(-i,p5),5+rempli);
affichage(translation(-i,c1),1+rempli);
affichage(translation(-i,c2),2+rempli);
affichage(translation(-i,c3),3+rempli);
affichage(translation(-i,c4),4+rempli);
d:=translation(-i+D-C,droite(K,M)):;
cc4:=symetrie(d,translation(-i,c4)):;
cc3:=symetrie(d,translation(-i,c3)):;
cc2:=symetrie(d,translation(-i,c2)):;
cc1:=symetrie(d,translation(-i,c1)):;
cc5:=symetrie(d,translation(-i,p5)):;
```

```
affichage(cc1,1+rempli);
affichage(cc2,2+rempli);
affichage(cc3,3+rempli);
affichage(cc4,4+rempli);
affichage(cc5,5+rempli);
hexagone(3-2*i,4-2*i);
R:=point(3-2*i+sqrt(3)*i):;;
affichage(translation(R-C,p3),3+rempli);
affichage(translation(R-C,p1),1+rempli);
affichage(translation(R-C,symetrie(J,p4)),4+rempli);
affichage(translation(R-C,p2),2+rempli);
affichage(translation(R-C,p5),5+rempli);
V:=point(3-2*i+i*sqrt(3)/2):;
W:=point(4-2*i+i*sqrt(3)/2):;
affichage(symetrie(droite(V,W),translation(R-C,p1)),1+rempli);
affichage(symetrie(droite(V,W),translation(R-C,p2)),2+rempli);
affichage(symetrie(droite(V,W),translation(R-C,p3)),3+rempli);
affichage(symetrie(droite(V,W),translation(R-C,symetrie(J,p4))),4+remp
affichage(symetrie(droite(V,W),translation(R-C,p5)),5+rempli);
```

On obtient :



ou bien en modifiant les définitions de `cc1..cc5` en remplacant les lignes :

```
d:=translation(-i+D-C,droite(K,M))):;
cc4:=symetrie(d,translation(-i,c4)):;
cc3:=symetrie(d,translation(-i,c3)):;
cc2:=symetrie(d,translation(-i,c2)):;
cc1:=symetrie(d,translation(-i,c1)):;
cc5:=symetrie(d,translation(-i,p5)):;
```

par :

```
S:=translation((D-C)-i,M):;
T:=translation(-i,symetrie(Q,N)):;
U:=translation(-i,N):;
d1:=mediatrice(T,S):;
cc4:=symetrie(d1,translation((T-U)-i,c4)):;
cc3:=symetrie(d1,translation((T-U)-i,c3)):;
cc2:=symetrie(d1,translation((T-U)-i,c2)):;
cc1:=symetrie(d1,translation((T-U)-i,c1)):;
cc5:=symetrie(d1,translation((T-U)-i,p5)):;
```

On obtient :



Chapitre 12

Pour s'amuser avec les probabilités

12.1 Les anniversaires de 3 personnes

Trois amis sont nès une même annèe de 365 jours.
On suppose que les dates de naissance sont équiprobables.
Quelles sont les probabilités pour que :

1/ Ils soient nés le même jour,

2/ Deux d'entre eux seulement soient nés le même jour,

3/ Ils soient nés a des dates différentes,

4/ Quelle relation vérifie ces trois réponses ?

Solution
On donne un numéro aux amis, i.e. on les ordonne.
Soit Ω l'univers ensemble de triplets de nombres allant de 1 à 365. Il y a 365^3 triplets possibles, donc $card(\Omega) = 365^3$.

1/ Soit A l'événement "Ils sont tous les trois nés le même jour". cela signifie que A est composé de triplés formés par 3 nombres égaux donc, $card(A) = 365$.
Donc, $p(A) = \frac{365}{365^3} = \frac{1}{365^2} \simeq 7.5e - 06$.

2/ Soit B l'événement "Deux seulement sont nés le même jour". B est composé de triplés formés par 2 nombres égaux et différents du 3-ième. Il y a trois possibilités (les deux premiers ou les deux derniers ou le premier et le troisième) ont le même anniversaire donc, comme il y a $365 * 364$ couples de nombres différents, $card(B) = 3 * 365 * 364$.
Donc, $p(B) = \frac{3*365*364}{365^3} = \frac{3*364}{365^2} \simeq 0.0082$.

3/ Soit C l'événement "Ils sont nés a des dates différentes". C est composé de triplés formés par 3 nombres différents donc, comme il y a $365 * 364 * 363$ triplets formés de 3 nombres différents, on a $card(C) = 365 * 364 * 363$.
Donc, $p(C) = \frac{365*364*363}{365^3} = \frac{364*363}{365^2} \simeq 0.99$.

4/ On doit avoir $p(A) + p(B) + p(C) = 1$ puisque A, B, C forment une partition de Ω. on a :
$1 + 3 * 364 + 364 * 363 = 1 + 364 * 366 = 1 + (365 - 1)(365 + 1) = 365^2$
donc on a bien : $p(A) + p(B) + p(C) = 1$.

12.2 Les anniversaires de `n` personnes

Dans une assemblée de `n` personnes, toutes sont nèes une annèe de 365 jours. On suppose que les dates de naissance sont équiprobables.
On note `p(n)` la probabilité pour que 2 personnes au moins aient leur anniversaire le même jour.

1/ Calculer `p(3)`,

2/ Donner la formule permettant de calculer `p(n)`,

3/ Déterminer une valeur approchée de `p(20)`, `p(30)` et `p(367)`,

4/ Déterminer le nombre `n` pour que l'on ait $\texttt{p(n)} \geq \frac{1}{2}$.

Solution
On donne un numéro aux personnes de l'assemblée, i.e. on les ordonne.
Soit Ω l'univers ensemble des `n`-uplets de nombres entiers de 1 à 365.
Il y a $\mathtt{365^n}$ triplets possibles, donc $\mathtt{card}(\Omega) = \mathtt{365^n}$.

1/ D'après l'exercice précédent, `p(3)` $=p(A)+p(B) = \frac{1+3*364}{365^2} \simeq 0.0082$.

2/ On va tout d'abord chercher la probabilité de l'événement contraire : soit D_n l'événement "Les `n` personnes ont leurs anniversaires a des dates toutes différentes".

Il y a $\mathtt{365} * \mathtt{364} * ... * \mathtt{365} - \mathtt{n} + \mathtt{1}$ triplets formés de nombres différents deux à deux, donc $card(D_n) = 365 * 364 * ...365 - n + 1 = \frac{365!}{(365-n)!}$.
Donc $p(D_n) = \dfrac{365!}{(365-n)! * 365^n} = \dfrac{364!}{(365-n)! * 365^{n-1}}$,
et donc $p(n) = 1-p(D_n) = \dfrac{(365-n)! * 365^n - 365!}{(365-n)! * 365^n} = \dfrac{(365-n)! * 365^{n-1} - 364!}{(365-n)! * 365^{n-1}}$.
Avec `Xcas` on peut définir `p(n)` on tape :
```
p(n):=1-factorial(364)/(factorial(365-n)*365^(n-1))
```
3/ Calculons avec `Xcas`, on tape :
```
evalf(perm(365,19)/365^19)
```
On obtient :
```
0.588561616419
```
donc $p(D_{20}) \simeq 0.588561616419$ et
`p(20)`$\simeq (1 - 0.588561616419) \simeq$`0.411438383581`
Ou on tape :
```
evalf(p(20))
```
On obtient :
```
0.411438383581
```
On tape :
```
evalf(perm(364,22)/365^22)
```
On obtient :
```
0.492702765676
```
donc $p(D_{23}) \simeq 0.492702765676$ et
`p(25)`$\simeq (1 - 0.492702765676) \simeq$`0.507297234324`
Ou on tape :
```
evalf(p(23))
```
On obtient :
```
0.507297234324
```
On tape :

`evalf(perm(365,29)/365^29)`
On obtient :
`0.293683757281`
donc $p(D_{30}) \simeq 0.293683757281$ et
`p(30)`$\simeq (1 - 0.293683757281) \simeq$`0.706316242719`
Ou on tape :
`evalf(p(30))`
On obtient :
`0.706316242719`
Ce qui veut dire que dans une assemblée de 20 personnes il y a 4 chances sur 10 pour que 2 personnes aient le même anniversaire, que dans une assemblée de 23 personnes il y a 1 chance sur 2 pour que 2 personnes aient le même anniversaire et que dans une assemblée de 30 personnes il y a 7 chances sur 10 pour que 2 personnes aient le même anniversaire ! ! ! !
Pour calculer $p(367)$, on n'a pas besoin de `Xcas` car :
$p(367) = 1$ puisqu'il n'y a que 365 dates possibles (ou 366 ...) parmi les 367 personnes et donc deux personnes au moins ont forcément le même anniversaire.

4/ On va utiliser le tableur pour chercher $p(20)..p(30)$, pour cela on tape dans `A0`..`A10` les valeurs de $p(D_{20})..p(D_{30})$:
`=evalf(perm(365,20+Row())/365^(20+Row()))`
et dans `B0` ..`B10` les valeurs de $p(20)..p(30)$:
`=1-A0`
Puis, on remplit les colonnes `A` et `B` avec ces formules à l'aide du bouton `remplir` (option `vers le bas`).
On rapelle que pour avoir la valeur d'une cellule dans la ligne de commande, on doit appuyer sur le bouton `eval`.
On obtient :
`B2=0.475695307663` et `B3=0.507297234324`
donc $n = 23$ car on a :
$p(23) = 0.507297234324 > 0.5$ et
pour $n = 22$ $p(22) = 0.475695307663 < 0.5$.
On peut aussi taper dans `C0` :
`=20+count_inf(0.5,B0:B10)`
car on sait que $20 < n < 30$ et que `coun_inf(0.5,B0:B10)` est la fonction qui compte le nombre d'éléments stricrement inférieurs à 0.5 dans la colonne `B` (de `B0` à `B10`).
On a mis 20 car il a `19+count_inf(0.5,B0:B10)` valeurs stricrement inférieures à 0.5 et donc `20+count_inf(0.5,B0:B10)` est la première valeur supérieure ou égale à 0.5.
On obtient dans `C0` :
`23`
On remarqera que :
`B21=0.903151611482`
Ce qui veut dire que dans une assemblée de 41 personnes il y a 9 chances sur 10 pour que 2 personnes aient le même anniversaire ! ! ! !

5/ On peut dessiner l'évolution des `p(n)` en fonction de `n` lorsque `n` varie entre 20 et 50. Il suffit poir cela de rajouter une colonne entre `A` et `B` on met `B0` en surbrillance et on appuie sur `c+`. La colonne `B` devient `C`, et une colonne `B` est

créée.
On tape alors `0` dans `B0`, puis dans `B1` on met `=B0+1` puis on remplit la colonne `B` avec cette formule.
Il suffit maintenant de mettre en surbrillance `B0:C30` puis d'ouvrir le menu `2d` et de choisir `Scatterplot` pour voir les différents points dans l'écran de géométrie (changer la configuration du graphique pour voir tous les points).

12.3 Les 4 dés du jeu de Win

On considère les 4 dés suivants :
— les faces du dé `A` ont comme points : `0,0,4,4,4,4`,
— les faces du dé `B` ont comme points : `1,1,1,5,5,5`,
— les faces du dé `C` ont comme points : `2,2,2,2,6,6`,
— les faces du dé `D` ont comme points : `3,3,3,3,3,3`,
La partie se compose de 12 lancers.
Pour une partie, chacun des joueurs choisit un dé. À chaque lancer, celui qui a le meilleur score marque 1 point. La partie se compose de 12 lancers.
On veut simuler ce jeu pour que l'on puisse jouer contre l'ordinateur. L'ordinateur tire au hasard un dé, vous donne son choix, puis vous choisisez un parmi les 3 dés qui restent. Puis vous jouez....
Quel dé faut-il choisir pour gagner contre l'ordinateur ?
On numérote les faces de chaque dé : par exemple, par ordre croissant des points des faces, ainsi pour le dé `A` les faces 0,1 ont comme points 0 et les faces 2,3,4,5 ont comme points 4. Pour jouer avec un dé, on tire au hasard un nombre entier entre 0 et 5 (`rand(6)`) pour voir sur quelle face tombe le dé, puis on regarde le nombre de points de cette face. Ce nombre dépend du dé choisi.
On écrit donc une fonction qui renvoie pour chaque dé la valeur de la face `n` du dé.

```
rande(des,n):={
if (des=="A"){if (n==0 or n==1) return 0 ;
                              else return 4;}};
if (des=="B"){if (n==0 or n==1 or n==2) return 1;
                              else return 5;}};
if (des=="C"){if (n==4 or n==5) return 6;
                              else return 2;}};
return 3;
}:;
```

Puis on écrit le programme qui correspond a une partie (12 lancers pour chacun) et qui renvoie la liste des scores (ordinateur,joueur).

```
jeuwin0():={
local deo,dem,po,pm,scoro,j;
deo:=char(rand(4)+65);
print("j'ai choisi le de "+ deo);
repeter saisir_chaine("votre choix",dem);
jusqua dem!=deo;
scoro:=0;
for (j:=0;j<12;j++){
```

```
po:=rande(deo,rand(6));
pm:=rande(dem,rand(6));
print(po,pm);
if (po>pm) scoro:=scoro+1;
}
return [scoro,12-scoro];
}
:;
```

On peut aussi utiliser les listes A,B,C,D pour representer chaque dé, et le programme devient beaucoup plus simple (on n'a pas besoin de la fonction `rande` ! ! ! !) mais il faut transformer le caractère contenu dans `deo` (par ex `"A"`) en `expr(deo)` (par ex en la valeur de la variable `A`). Pour `dem` on ne saisit plus une chaîne mais directement le nom d'une variable avec `saisir(dem)` au lieu de `saisir_chaine(dem)`.

```
jeuwin():={
local deo,dem,po,pm,scoro,j,A,B,C,D;
deo:=char(rand(4)+65);
print("j'ai choisi le de "+ deo);
A:=[0,0,4,4,4,4];
B:=[1,1,1,5,5,5];
C:=[2,2,2,2,6,6];
D:=[3,3,3,3,3,3];
deo:=expr(deo);
repeter saisir("votre choix",dem);
jusqua dem!=deo;
scoro:=0;
for (j:=0;j<12;j++){
po:=deo[rand(6)];
pm:=dem[rand(6)];
print(po,pm);
if (po>pm) scoro:=scoro+1;
}
return [scoro,12-scoro];
}
:;
```

On peut ensuite faire plusieurs parties. On tape :
`score:=[0,0];`
puis par exemple
`score:=score+jewin()`
plusieurs fois et on obtient les scores cumulés.

Pour savoir quel dé il faut choisir, on cherche la probabilité que l'ordinateur gagne selon les différents choix :

1. l'ordinateur a choisi le dé `A`,
   — si vous choisissez le dé `B`,
     L'ordinateur gagne si le dé `A` fait quatre et le dé `B` fait un, donc

$$P(nA > nB) = (4/6) * (3/6) = 1/3$$

— si vous choisissez le dé `C`,
L'ordinateur gagne si le dé `A` fait quatre et le dé `C` fait deux, donc

$$P(nA > nC) = (4/6) * (4/6) = 4/9$$

— si vous choisissez le dé `D`,
L'ordinateur gagne si le dé `A` fait quatre, donc

$$P(nA > nD) = (4/6) = 2/3$$

Il faut donc choisir le dé `B` pour gagner avec une probabilité de 2/3.

2. l'ordinateur a choisi le dé `B`,
— si vous choisissez le dé `C`,
L'ordinateur gagne si le dé `B` fait cinq et le dé `C` fait deux, donc

$$P(nB > nC) = (3/6) * (4/6) = 1/3$$

— si vous choisissez le dé `D`,
L'ordinateur gagne si le dé `B` fait cinq, donc

$$P(nB > nD) = (3/6) = 1/2$$

— si vous choisissez le dé `A`,
cas déjà vu : $P(nB > nA) = 1 - P(nA > nB) = 2/3$
Il faut donc choisir le dé `C` pour gagner avec une probabilité de 2/3.

3. l'ordinateur a choisi le dé `C`,
— si vous choisissez le dé `D`,
L'ordinateur gagne si le dé `C` fait six, donc

$$P(nC > nD) = (2/6) = 1/3$$

— si vous choisissez le dé `B`,
cas déjà vu : $P(nC > nB) = 1 - P(nB > nC) = 2/3$
— si vous choisissez le dé `A`,
cas déjà vu : $P(nC > nA) = 1 - P(nA > nC) = 5/9$
Il faut donc choisir le dé `D` pour gagner avec une probabilité de 2/3.

4. l'ordinateur a choisi le dé `D`,
— si vous choisissez le dé `A`,
cas déjà vu : $P(nD > nA) = 1 - P(nA > nD) = 1/3$
— si vous choisissez le dé `B`,
cas déjà vu : $P(nD > nB) = 1 - P(nB > nD) = 1/2$
— si vous choisissez le dé `C`,
cas déjà vu : $P(nD > nC) = 1 - P(nC > nD) = 2/3$
Il faut donc choisir le dé `A` pour gagner avec une probabilité de 2/3.

Remarque
La relation "le dé N_1 gagne le dé N_2" n'est pas transitive, en effet :
le dé `B` gagne le dé `A`,
le dé `C` gagne le dé `B`,
le dé `D` gagne le dé `C`,
le dé `A` gagne le dé `D`,

De plus, le jeu est trompeur car le choix ne depend pas du score moyen de chaque dé, en effet :
le dé `A` fait en moyenne un score de 8/3,
le dé `B` fait en moyenne un score de 3,
le dé `C` fait en moyenne un score de 10/3,
le dé `D` fait en moyenne un score de 3 et pourtant le dé `D` l'emporte sur le dé `C` de moyenne 10/3 mais il perd contre le dé `A` qui n'a qu'une moyenne de 8/3 !

12.4 Des calculs de moyenne

12.4.1 Nombre d'enfants moyen par famille

Dans un pays, le roi a décidé que les familles de ses sujets doivent avoir des enfants jusqu'à ce qu'elles aient un garçon.
Quelle est le nombre d'enfants moyen par famille ?

Si X est la variable aléatoire égale au nombre d'enfants dans une famille, on a :

— $P(X = 1) = \frac{1}{2}$
— $P(X = 2) = \frac{1}{2^2}$
— $P(X = 3) = \frac{1}{2^3}$
— ....
— $P(X = k) = \frac{1}{2^k}$

Donc $E(X) = \sum_{k=1}^{+\infty} k * \frac{1}{2^k}$
On tape :
`sum(k/2^k,k,1,+infinity)`
On obtient :
`2`
Donc le nombre moyen d'enfants est 2....On aurait pu s'en douter car dans chaque famille il n'y a qu'un seul garçon et comme en moyenne il nait autant de filles que de garçons, il y aura en moyenne autant de filles que de garçons, soit 2 enfants en moyenne dans chaque famille.

12.4.2 Nombres triangulaires aléatoires

On tire au hasard des nombres entre 1 et n jusqu'à obtenir 1. Le résultat est alors la somme des nombres obtenus. Quel est la moyenne des résultats obtenus ?

La solution mathématique

Supposons pour commencer $n = 2$
Les résultats peuvent être : $R = 1, 3, 5....2p + 1....$
On a :
$P(R_2 = 1) = \frac{1}{2}$,
$P(R_2 = 3) = \frac{1}{2^2}$

.... $P(R_2 = 2p+1) = \dfrac{1}{2^{p+1}}$
Donc :
$E(R_2) = \displaystyle\sum_{p=0}^{+\infty}(2p+1)\frac{1}{2^{p+1}}$
On tape :
```
sum((2k+1)/2^(k+1),k,0,+infinity)
```
On obtient :
```
3
```
La moyenne de R_2 vaut donc 1+2=3.

Peut-on généraliser ?
Dans le cas général, on tire au hasard des nombres entre 1 et n jusqu'à obtenir 1. La moyenne des sommes des nombres tirés vaut-elle $1+2+...+n = n(n+1)/2$?
Soit X_n la variable aléatoire égale au nombre de tirages parmi $1...n$ qu'il faut effectuer pour obtenir 1.
On a :
$P(X_n = 1) = \dfrac{1}{n}$,
$P(X_n = 2) = \dfrac{1}{n^2}$ et les résultats obtenus peuvent être :
$2+1 = 3, 3+1 = 4, ..., n+1$ qui est une liste L_2 de taille $n-1$ et de somme :
$2+3+...n+n-1 = (n-1)(n+3)/2$
$P(X_n = 3) = \dfrac{1}{n^3}$ et les résultats obtenus peuvent être :
$2+2+1 = 5, 2+3+1 = 6, 3+2+1 = 6, ...n+n+1$ qui est une liste L_3 de taille $(n-1)^2$
Que vaut la somme de cette liste ?
Chaque terme est la somme de 2 termes et de 1 : dans ces sommes chaque nombre (2,3,...n) apparaissent autant de fois donc il y a $2(n-1)^2/(n-1) = 2n-2$ fois 2, $2n-2$ fois 3...$2n-2$ fois n et $(n-1)^2$ fois 1. La somme cette liste vaut donc :
$(n-1)^2+(2n-2)(2+3+...+n) = (n-1)^2+(n-1)^2(n+2) = (n-1)^2(n+3)$.
...... $P(X_n = p) = \dfrac{1}{n^p}$ et les résultats obtenus peuvent être : $2+...+2+1 = 2p-1, 2+...+3+1 = 2p, 3+2+...+2+1 = 2p, ...$ (liste L_p de taille $(n-1)^{p-1}$)
Que vaut la somme de cette liste ?
Cette somme est composée de $p*(n-1)^{p-1}$ termes.
Cette somme est la somme :
de $(n-1)^{p-1}$ fois 1,
de $(p-1)(n-1)^{p-2}$ fois 2
....
de $(p-1)(n-1)^{p-2}$ fois n
donc elle vaut :
$(n-1)^{p-1}+(p-1)(n-1)^{p-2}(2+3+...+n) =$
$(n-1)^{p-1}+(n-1)^{p-1}(p-1)(n+2)/2 = (n-1)^{p-1}(1+(p-1)(n+2)/2) =$
$(n-1)^{p-1}((p(n+2)-n)/2)$.
Donc :
$E(R_n) = \sum_{p=1}^{+\infty}\frac{1}{n^p}(n-1)^{p-1}(p(n+2)-n)/2$
On tape :
```
sum((n-1)^(p-1)/n^p*(p*(n+2)-n)/2 ,p,1,+infinity)
```

On obtient :
`(n^2+n)/2`
Donc la moyenne de R_n est égale à $1 + 2 + ...n$

La modélisation avec `Xcas`

On tape le programme `trialea(r,q,p)` qui tire au hasard des nombres entre 1 et r. On fait p fois des échantillons de taille q, et on dessine les résultats intermédiaires obtenus : `l` contient les sommes cumulées des résultats (ici une somme) c'est à dire la somme d'un échantillon de taille `n=k+1+j*q` avec `k=0..q-1` et `j=0..p-1`. Dans `Ldiv` on met `evalf(l/n)` lorsque `n=q,2*q...p*q`

```
trialea(r,q,p):={
  local j,k,l,n,LdivN,alea;
  LdivN:=NULL;
  l:=0;
  n:=0;
  for (j:=0;j<p;j++){
    for (k:=0;k<q;k++){
      alea:=(rand(r)+1);
      while (alea!=1){
        l:=l+alea;
        alea:=(rand(r)+1);
      }
      l:=l+1;
      n:=n+1;
    }
    LdivN:=LdivN,evalf(l/n);
  }
  return LdivN;
}:;
```

On tape :
`L10:=trialea(10,100,10000);`
`plotlist(L10)`
On obtient :



12.4.3 Factorielles aléatoires

On tire au hasard des nombres entre 1 et n jusqu'à obtenir 1. Le résultat est alors le produits des nombres obtenus. Quel est la moyenne des résultats obtenus ?

La solution mathématique

Cela ressemble à l'exercice précédent.....
Supposons pour commencer $n = 2$.
Les résultats peuvent être : $R = 1, 2, 4....2^p...$
On a :
$P(R_2 = 1) = \frac{1}{2}$,
$P(R_2 = 2) = \frac{1}{2^2}$
.... $P(R_2 = 2^p) = \frac{1}{2^{p+1}}$
Donc :
$E(R_2) = \sum_{p=0}^{+\infty} 2^p \frac{1}{2^{p+1}}$ On tape :
`sum(2^k/2^(k+1),k,0,+infinity)`
On obtient :
`infinity`
La moyenne de R_2 est donc infinie.

Peut-on généraliser ? Dans le cas général, on tire au hasard des nombres entre 1 et n jusqu'à obtenir 1. La moyenne des produits des nombres tirés est-elle infinie ?
Soit X_n la variable aléatoire égale au nombre p de tirages parmi $1...n$ qu'il faut effectuer pour obtenir 1.
On a :
$P(X_n = 1) = \frac{1}{n}$,
$P(X_n = 2) = \frac{1}{n^2}$ et les résultats obtenus peuvent être :
$2*1 = 2, 3*1 = 3, ..., n*1$ (liste L_2 de taille $n-1$ de produit $2*3+...*n = n!$)
$P(X_n = 3) = \frac{1}{n^3}$ et les résultats obtenus peuvent être :
$2*2*1 = 4, 2*3*1 = 6, 3*2*1 = 6, ...n*n*1$ (liste L_3 de taille $(n-1)^2$)
Que vaut la somme de cette liste ?
Chaque terme de cette liste provient du developpement de :
$(2+3+...+n)^2$ donc la somme de la liste L_3 vaut $(2+3+...+n)^2$
.....
$P(X_n = p) = \frac{1}{n^p}$ et les résultats obtenus peuvent être :
$2*...*2*1 = 2^{p-1}, 2+...+3+1 = 2p-2*3, , ...$ (liste L_p de taille $(n-1)^{p-1}$)
Que vaut la somme de cette liste ?
Chaque terme de cette liste provient du developpement de :
$(2+3+...+n)^{p-1}$ donc la somme de la liste L_p vaut $(2+3+...+n)^{p-1}$
Donc :

$$E(R_n) = \sum_{p=1}^{+\infty} \frac{1}{n^p}(2+3+...+n)^{p-1} = \frac{1}{n}\sum_{p=1}^{+\infty}((n+2)*(n-1)/(2*n))^{p-1}$$

On obtient une somme géométrique de raison $(n+2)*(n-1)/(2*n) >= 1$ pour $n >= 2$.
On tape :
`sum(((2+n)*(n-1)/2)^p-1/(n)^p ,p,1,k)`
On obtient :
`infinity`
Donc la moyenne de R_n est infinie.

La modélisation avec `Xcas`

On tape le programme `factalea(r,q,p)` qui tire au hasard des nombres entre 1 et r. On fait p fois des échantillons de taille q, et on dessine les résultats intermédiaires obtenus : `l` contient les sommes cumulées des résultats (ici un produit) c'est à dire la somme d'un échantillon de taille `n=k+1+j*q` avec `k=0..q-1` et `j=0..p-1`. Dans `Ldiv` on met `evalf(l/n)` lorsque `n=q,2*q...p*q`

```
factalea(r,q,p):={
  local j,k,l,n,LdivN,alea;
  LdivN:=NULL;
  l:=0;
  n:=0;
  for (j:=0;j<p;j++){
    for (k:=0;k<q;k++){
      alea:=(rand(r)+1);
      f:=1
      while (alea!=1){
        f:=f*alea;
        alea:=(rand(r)+1);
      }
      l:=l+f;
      n:=n+1;
    }
    LdivN:=LdivN,evalf(l/n);
  }
  return [LdivN];
}:;
```

On tape :
`F3:=factalea(3,10,100);`
`plotlist(F3)`
On obtient :


y
2.5e+05
2e+05
1.5e+05
1e+05
50000
0
0
20
40
60
80
100

Chapitre 13

Pour s'amuser avec les séries

Soit n un entier positif.
Soit c_n le nombre de triplets (X, Y, Z) de $\mathbb{N}$ qui vérifient :

$$X + 2Y + 4Z = n$$

On veut calculer c_n.
Déterminer c_{100} et c_{1000}.

On propose pour cela la téchnique suivante :
- Effectuer un développement en série au voisinage de l'origine de :
$f_1(x) = \dfrac{1}{1-x}$,
$f_2(x) = \dfrac{1}{1-x^2}$,
$f_3(x) = \dfrac{1}{1-x^4}$,
$f(x) = \dfrac{1}{(1-x)(1-x^2)(1-x^4)}$,
- Montrer, en effectuant le produit des 3 développements en série de f_1, f_2, f_3, que le coefficient de x^n du développement de f est c_n.
- Déterminer le développement de f par une autre méthode.
- En déduire c_n.

On tape :
$\texttt{series}(\dfrac{1}{(1-\texttt{x})(1-\texttt{x}^2)(1-\texttt{x}^4)}, 0, 20)$
On obtient :
$1 + \texttt{x} + 2 * \texttt{x}^2 + 2 * \texttt{x}^3 + 4 * \texttt{x}^4 + 4 * \texttt{x}^5 + 6 * \texttt{x}^6 + 6 * \texttt{x}^7 + 9 * \texttt{x}^8 + 9 * \texttt{x}^9 + 12 * \texttt{x}^{10} + 12 * \texttt{x}^{11} + 16 * \texttt{x}^{12} + 16 * \texttt{x}^{13} + 20 * \texttt{x}^{14} + 20 * \texttt{x}^{15} + 25 * \texttt{x}^{16} + 25 * \texttt{x}^{17} + 30 * \texttt{x}^{18} + 30 * \texttt{x}^{19} + 36 * \texttt{x}^{20} + \texttt{x}^{21} * \texttt{order\_size}(\texttt{x})$
On remarque que les coefficients sont :
$1, 1, 2, 2, 4, 4, 6, 6, 9, 9, 12, 12, 16, 16, 20, 20, 25, 25, 30, 30, 36$...
On obtient les carrés des entiers puis, le produit de 2 entiers consécutifs :
$1, 4, 9, 16, 25, 36$ et $1 * 2, 2 * 3, 3 * 4, 4 * 5, 5 * 6$...
On suppose donc que :
$f(x) = \sum_{n=0}^{\infty} c_n x^n$ avec :
$c_{4*k} = c_{4*k+1} = (k+1)^2$ et

$c_{4*k+2} = c_{4*k+3} = (k+1)*(k+2)$
ce qui donne bien $c_0 = c_1 = 1$, $c_2 = c_3 = 2$, $c_4 = c_5 = 4$, $c_6 = c_7 = 6$...
On a donc :
$c_{100} = c_{4*25} = 26^2 = 676$
$c_{1000} = c_{4*250} = 251^2 = 63001$
On peut bien sûr le vérifier en demandant à `Xcas` :
$\texttt{series}(\frac{1}{(1-x)(1-x^2)(1-x^4)}, 0, 100)$ et
$\texttt{series}(\frac{1}{(1-x)(1-x^2)(1-x^4)}, 0, 1000)$

Mais comment montrer que l'on a bien :
$f(x) = \sum_{n=0}^{\infty} c_n x^n$ avec :
$c_{4*k} = c_{4*k+1} = (k+1)^2$ et
$c_{4*k+2} = c_{4*k+3} = (k+1)*(k+2)$
On peut penser à décomposer la fraction rationnelle f :
On tape :
$\texttt{partfrac}(\frac{1}{(1-x)(1-x^2)(1-x^4)})$
On obtient :
$\frac{x+1}{8*(x^2+1)} + \frac{5}{32*(x+1)} + \frac{1}{16*(x+1)^2} - \frac{9}{32*(x-1)} + \frac{1}{4*(x-1)^2} - \frac{1}{8*(x-1)^3}$
ce qui n'est pas simple....

Pour le montrer on peut commencer par montrer que :
$\texttt{series}(\frac{1}{(1-x^2)(1-x^4)}, 0, 20)$ vaut :
$1 + x^2 + 2*x^4 + 2*x^6 + 3*x^8 + 3*x^{10} + 4*x^{12} + 4*x^{14} + 5*x^{16} +$
$5*x^{18} + 6*x^{20} + x^{21}*\texttt{order\_size}(x)$ c'est à dire :
$\texttt{series}(\frac{1}{(1-x^2)(1-x^4)}, 0, 20) = \sum_{n=0}^{\infty} c_n x^n$ avec :
$c_{4*k} = c_{4*k+2} = (k+1)$ et
$c_{4*k+1} = c_{4*k+3} = 0$
puis multiplier par cette série par $\sum_{n=0}^{\infty} x^n$
On peut aussi regarder le développement en série de $f/(1+x)$ car :
$\frac{f}{1+x} = \frac{1}{(1-x^2)^2(1-x^4)}$.
On a ;
$1/(1-x^2)^2 = \sum_{n=0}^{\infty}(n+1)x^{2n}$ $(1/(1-u)^2 = (1/(1-u))'$ puis $u = x^2)$
$1/(1-x^4) = \sum_{n=0}^{\infty} x^{4n}$
on multiplie ces deux séries et on obtient :
coefficient de x^{4n} : $1 + 3 + 5 + ... + (2n+1) = (n+1)^2$
coefficient de x^{4n+2} : $2 + 4 + 6 + ... + (2n+2) = (n+1)(n+2)$
donc
$f = (1+x)\sum_{n=0}^{\infty}(n+1)^2 x^{4n} + (n+1)(n+2)x^{4n+2}$
ce qui donne bien la formule annoncée.

Vous pouvez maintenant vous amuser avec le problème similaire :
Soit n un entier positif. Soit c_n le nombre de triplets de $\mathbb{N}$ qui vérifient :

$$X + 2Y + 5Z = n$$

Déterminer c_{100} et c_{1000} en calculant c_n.

Chapitre 14

Pour s'amuser en géométrie plane

14.1 Des problèmes de plus court trajet

Les problème de plus court trajet sont souvent difficiles... En voici quelques uns plutôt faciles.

14.1.1 Comment placer un pont

Deux villages assimilés à deux points A et B sont situés de part et d'autre d'une rivière assimilée à deux droites parallèles $D1$ et $D2$. Où doit-on placer un pont PQ (perpendiculairement aux berges) sur la rivière pour minimiser le trajet allant de A à B ?
On veut que le trajet $AP+PQ+QB$ soit minimum, on remarque que dans le trajet PQ est constant et est égal à la largeur de la rivière. On dessine le parallélogramme $APQR$ et ainsi, $AP + PQ = AR + RQ$. On a donc :
$AP + PQ + QB = AR + RQ + QB$ où $AR = PQ$=cste
La solution est maintenanant évidente : pour rendre minimum $RQ + QB$ il suffit de choisir A, Q, B alignés.
Le dessin avec `Xcas` :
On clique deux points `A` à gauche de $x = -1$ et `B` à droite de $x = 1$. On peut ensuite faire bouger les points `A` ou `B` et visualiser les trajets `APQB` et `ARQB`. On tape par exemple :

```
A:=point([-5/2,1,'affichage'=0]);
B:=point([5,-2,'affichage'=0]);
D1:=droite(-1,-1+i):;D1;
D2:=droite(1,1+i):;D2;
R:=translation(2,A);
polygone_ouvert(A,R,Q,affichage=1);
segment(R,B,affichage=ligne_tiret_point+ 4);
d:=mediatrice(R,B);
Q:=inter_unique(d,D2);
P:=translation(-2,Q);
polygone_ouvert(A,P,Q,B);
supposons(a=[-2.86,-5,5,0.01]);
p:=point(-1+i*a);
```

```
q:=point(1+i*a);
polygone_ouvert(A,p,q,B);
segment(R,q,affichage=1);
legende(-6+4i,"Ap="+string(evalf(longueur(A,p))));
legende(-6+3i,"Bq="+string(evalf(longueur(B,q))));
```

On peut ensuite faire bouger les points `p` et `q` et visualiser les trajets `ApqB` et `ARqB`.
On obtient :

14.1.2 Comment placer deux ponts

Deux villages assimilés à deux points A et B sont situés de part et d'autre de deux rivières, l'une est assimilée à deux droites parallèles $D1$ et $D2$ et l'autre est assimilée à deux droites parallèles $D3$ et $D4$.
Où doit-on placer deux ponts $P1P2$ et $P3P4$ sur les rivières (perpendiculairement aux berges) pour minimiser le trajet allant de A à B.
?
On fait le dessin avec `Xcas` :
On clique deux points `A` en bas à gauche de l'écran et `B` en haut et à droite de l'écran et on tape :

```
assume(a:=1);
D1:=droite(-2,-2+i);
D2:=droite(-1,-1+i);
D3:=droite(-1,a-1-i);
D4:=droite(0,a-i);
R:=translation(1,A);
segment(A,R);
Q:=translation(-(1+a*i)/(1+a^2),B);
```

```
segment(B,Q);
P2:=inter(droite(R,Q),D2)[0];
P1:=translation(-1,P2);
P3:=inter(droite(R,Q),D3)[0];
P4:=translation((1+a*i)/(1+a^2),P3);
segment(A,P1);
segment(P1,P2);
segment(P2,P3);
segment(P3,P4);
segment(P4,B);
segment(R,P2);
segment(P3,Q);
```

Il reste à observer le dessin en faisant bouger `a` ou `A` ou `B` pour voir que :
$AR = AP1$ =largeur d'une rivière
$BQ = BP4$ =largeur de l'autre rivière
$AP1 + P1P2 + P2P3 + P3P4 + P4B = AR + RP2 + P2P3 + P3Q + QB = AR + RQ + QB$
et comprendre comment on fait la construction des deux ponts.

14.1.3 Minimiser AMB avec M sur une droite

Soient une droite `d` et deux points `A` et `B`. On veut minimiser la distance `AM+MB` lorsque `M∈d`.
Si les deux points sont de part et d'autre de `d`, c'est facile on trace la droite `AB`,
si les deux points sont situés dans le même demi-plan défini par `d`, on se raméne à la situation précédente en prenant le symétrique `C` de `B` par rapport à `d`. Ainsi, `AM+MB=AM+MC` et `A` et `C` sont de part et d'autre de `d`. Le dessin avec `Xcas` :
On clique deux points `A` et `B` à droite de $x = -1$.

```
d:=droite(-1,-1+i);
C:=symetrie(d,B);
M:=inter(droite(A,C),d)[0];
segment(A,M);
segment(M,B);
segment(C,M);
N:=element(d);
segment(A,N);
segment(N,B);
segment(C,N);
```

On peut ensuite faire bouger les points `N` ou `B` et visualiser les trajets `AMB` et `AMC` en les comparant à `ANB` et `ANC`.

14.1.4 Minimiser $AMNB$ avec M et N chacun sur une droite

Soient deux droites `d1, d2` et deux points `A` et `B`. On veut minimiser la distance `AM+MN+NB` lorsque `M∈d1` et `N∈d2`.
Les deux droites définissent quatre portions de plan (I,II,III,IV) (I et III étant opposés par le sommet). Il y a plusieurs cas à distinguer et selon la position de `A` et

B par rapport à ces portions de plan. Selon les cas, pour trouver la solution il faut tracer le symétrique A1 de A par rapport à d1 et le symétrique B2 de B par rapport à d2, puis tracer soit AB, soit AB2, soit A1B, soit A1B2.

14.1.5 Minimiser $AMNB$ avec M et N sur une droite d en imposant $MN = L$

Ici le vecteur $\overrightarrow{MN}$ est constant car il est parallèle à d il est de longueur constante L est a la même direction que le vecteur $\overrightarrow{ab}$ où a et b sont les projection orthogonales de A et B sur d.
Soit R le translaté de A par le vecteur $\overrightarrow{MN}$. On a donc $AMNR$ est un parallélogramme et $AM = RN$ et $AR = MN$. Le trajet a minimiser est donc : $AM + MN + NB = RN + AR + NB = AR + RN + NB$.
Puisque A et R sont fixes il faut minimiser $RN + NB$.
Deux cas de figures :

- B et A sont de part et d'autres de d. Il suffit de choisir N comme intersection de RB et de d.
- B et A sont d'un même côté de d. Il suffit de construire le symétrique $B1$ de B par rapport à d, puis de choisir N comme intersection de $RB1$ et de d.

14.1.6 Un trajet difficile : minimiser AMB avec M sur un cercle

Soient deux points A et B.
Un point M se déplace sur le cercle C de centre O et de rayon 1. On choisit A et B pour que la droite AB ne coupe pas le cercle C.
On cherche dans ce cas, à minimiser le trajet AMB.
On fait une simulation avec Xcas
On va faire apparaître sur le même écran, le dessin géométrique et le graphe de la fonction longueur(AM)+longueur(MB)-2 lorsque M varie (on enlève 2 pour pouvoir voir le graphe en entier).
On régle la fenêtre graphique pour voir :
$[-3.5, 6.5] \times [-1, 4.4]$
On clique sur deux points pour définir A et B (par exemple A:=point(-3,-2);B:=point(1,-2);)
On tape :

```
A:=point(-3,-2);
B:=point(1,-2);
C:=cercle(0,1);
t:=element(0..2*pi);
M:=point(exp(i*t)); // ou M:=element(C,t);
L(A,B,t):=evalf(longueur(A,exp(i*t))+longueur(B,exp(i*t)));
G:=plotfunc(L(A,B,x)-2,x);
N:=element(G,t);
bissectrice(M,A,B);
exbissectrice(M,A,B)
```

Ensuite lorsque l'on fait bouger t les points M et N bougent, l'un sur le cercle C, l'autre sur le graphe G et l'on peut voir que le minimum est atteint quand la bissec-

trice intérieure de l'angle $\widehat{\texttt{AMB}}$ passe par O.
On obtient :



On peut aussi faire varier B pour voir ce qu'il se passe quand la droite AB coupe C c'est à dire quand la solution est evidente...

Cas particulier

On peut démontrer que lorsque le triangle OAB est isocéle de sommet O le point M du cercle C qui rend le trajet AM+MB minimum se trouve sur la bissectrice intérieure de l'angle $\widehat{\texttt{AMB}}$. En effet soit deux points N1 et N2 du cercle C symétriques par rapport à cette bissectrice (qui est aussi la médiatrice de AB). On a donc AN1=BN2 et AN2=BN1 et donc AN1+N1B=AN1+AN2.
Soient I le milieu de N1N2 et J le milieu de AB. Les points O, I, M, J sont tous sur la médiatrice de AB et puisque JI>JM (I milieu de la corde N1N2 et J milieu de l'arc N1N2) et on en déduit que AI>AM.
Puisque $\overrightarrow{AN1}+\overrightarrow{AN2}=2\overrightarrow{AI}$, d'aprés l'inégalité triangulaire on a 2AI<AN1+AN2.
Donc AM+MB=2AM<2AI<AN1+AN2 ce qui prouve que AM+MB est minimum.

Cas général

Soient un cercle C de centre O, une droite d extéreure au cercle C et 2 points A et B sur la droite d. On cherche un point M sur le cercle C pour que AM+BM soit minimum.
On sait que le lieu des points M tel que MA+MB=2a (pour a constant) est une ellipse de foyers A et B dont la normale en M est la bissectrice intérieure de l'angle $\widehat{\texttt{AMB}}$.
Soit M sur le cercle C, MA+MB sera minimum lorsque l'ellipse de foyers A et B et passant par M sera tangente au cercle C, c'est à dire lorsque l'ellipse et le cercle C auront la même normale au point M. La normale en M au cercle C de centre O est OM, la normale en M à l'ellipse de foyer A et B est la bissectrice intérieure de l'angle $\widehat{\texttt{AMB}}$. Donc l'ellipse sera tangente au cercle quand la bissectrice intérieure de l'angle $\widehat{\texttt{AMB}}$ passe par O.
Lorsque la bissectrice intérieure de l'angle $\widehat{\texttt{AMB}}$ passe par O, son intersection avec le cercle donne la position de 2 points M qui donneront le maximum et le minimum

du trajet MA+MB.
La construction de M n'est pas facile. On choisit un système de coordonnées pour avoir M sur le cercle de centre O et de rayon 1 et AB parallèle à l'axe des x.
On tape par exemple :

```
supposons(a=[2.0,-5,5,0.1]);
supposons(c=[0.6,-3.2,3.2,0.1]);
O:=point(0);
M:=point(exp(i*c));
cercle(0,1);
A:=point(2,a);
B:=point(5,a);
segment(M,A);
segment(M,B);
d:=bissectrice(M,A,B);
```

En faisant bouger c pour que d passe par O, on obtient pour a=2 : c=0.6.





On tape :
`Eq:=simplify(equation(d))`
`fsolve(subst(Eq,[x,y],[0,0]),c,0` On obtient comme valeur de `c` pour `a=2` :
`0.605888470356`

14.1.7 Encore un trajet à minimiser MA+MB+MC

Soient trois points `A,B,C` dans le plan. On cherche le point `M` qui minimise le trajet`MA+MB+MC`.
On sait que le lieu des points `M` tel que `MA+MB=2a` (pour `a` constant) est une ellipse de foyers `A` et `B` dont la normale en `M` est la bissectrice intérieure de l'angle $\widehat{\texttt{AMB}}$.
Si le point `M` réalise le minimum, `MA+MB+MC` sera minimum lorsque `C` sera situé sur la bissectrice intérieure de l'angle $\widehat{\texttt{AMB}}$ puisque cette bissectrice est la normale à l'ellipse de foyers `A` et `B` et passant par `M`.
Donc $\widehat{\texttt{BMC}} = \widehat{\texttt{CMA}}$
De même, si le point `M` réalise le minimum, `MA+MB+MC` sera minimum lorsque `A` sera situé sur la bissectrice intérieure de l'angle $\widehat{\texttt{BMC}}$ puisque cette bissectrice est la normale à l'ellipse de foyers `B` et `C` et passant par `M`.
Donc $\widehat{\texttt{CMA}} = \widehat{\texttt{AMB}}$
Si `M` est à l'intérieur du triangle `ABC`, on en déduit que :
$\widehat{\texttt{BMC}} = \widehat{\texttt{CMA}} = \widehat{\texttt{AMB}} = 2 * \pi/3$



Pour qu'un tel point existe il faut que le triangle ABC ne possède pas un angle de mesure de plus de $2 * \pi/3$ radians.
On fait une simulation avec `Xcas`
On fait le dessin en tapant par exemple :

```
A:=point([-3,-2,'affichage'=0]);
B:=point([1,-2,'affichage'=0]);
```

```
a:=element((-5) .. 5,0.8,0.1);
b:=element((-5) .. 5,-1.7,0.1);
c:=element(0 .. 5,2.8,0.1);
C:=point(3,c);
M:=point(a,b);
L(A,B,C,a,b):=evalf(longueur(A,point(a,b))+
      longueur(B,point(a,b))+longueur(C,point(a,b)));
G:=plotfunc(L(A,B,C,x,b)-8,x);
N:=element(G,a);
bissectrice(M,A,B);
exbissectrice(M,A,B);
droite(B,B+exp(i*pi/3));
arc(B,A,2*pi/3);
arc(C,B,2*pi/3);
arc(A,C,2*pi/3);
```

On obtient pour (a,b,c)=(0.8,1.7,2.8) :



La solution

Soient `A:=point(xa,ya):;B:=point(xb,yb):;C:=point(xc,yc):;`, `M:=point(x,y):;` et la fonction f de $\mathbb{R}^2$ dans $\mathbb{R}$ qui à x, y fait correspondre $MA + MB + MC$.

On a :

`f(x,y):=longueur(M,A)+longueur(M,B)+longueur(M,C)`.

Cherchons les points critiques de f c'est à dire les points où son gradient : `grad(f(x,y),[x,y])` s'annule.

On tape pour avoir le gradient de `longueur(M,A)` lorsque $M! = A$:

`grad(longueur(M,A)),[x,y])`

On obtient :

`[(x-xa)/sqrt((x-xa)^2+(y-ya)^2),(y-xa)/sqrt((x-xa)^2+(y-ya)^2)]`

c'est à dire le vecteur $\overrightarrow{AM}/|AM|$ qui est norme 1.
Donc si $M! = A$ et $M! = B$ et $M! = C$, le gradient de f existe et est égal à est la somme de 3 vecteurs de norme 1.
Ce gradient s'annule si :
$\overrightarrow{MA}/|MA| + \overrightarrow{MB}/|MB| + \overrightarrow{MC}/|MC| = 0$
c'est à dire lorsque les vecteurs $\overrightarrow{MA}, \overrightarrow{MB}, \overrightarrow{MC}$ font entre eux un angle de $2\pi/3$.
On en déduit que si les angles du triangle sont inférieurs à $2\pi/3$, le point M qui réalise le minimum de $MA + MB + MC$ est l'intersection des arcs capables AB, AC, BC d'angle $2\pi/3$.
On tape :

```
B:=point(0);
C:=point(2);
A:=point(5/2+i*4);
triangle(A,B,C);
g1:=arc(B,A,2*pi/3):;g1;
g2:=arc(C,B,2*pi/3):;g2;
g3:=arc(A,C,2*pi/3):;g3;
M:=inter_unique(g1,g2);
```

On obtient :



Si un des angles du triangle est supérieur à $2\pi/3$, les arcs `g1,g2,g3` ne sont pas concourants et donc le point M qui réalise le minimum de $MA + MB + MC$ se trouve soit en A, soit en B soit en C.
En effet lorsque le point M réalise le minimum de $MA + MB + MC$ c'est que M est soit un point critique de f, lorsque f est dérivable, soit un point où f n'est pas dérivable.
Supposons que l'angle $c = (\overrightarrow{CA}, \overrightarrow{CB})$ du triangle ABC soit supérieur à $2\pi/3$.
On a $\cos(c) < 0$ et $AB^2 = AC^2 + BC^2 - 2 * AC * BC * \cos(c)$ donc $AB > AC$

et $AB > BC$ donc $BC + AC < BC + AB$ et $BC + AC < AB + AC$
Donc le point qui réalise le minimum se trouve en C.



14.2 Des problèmes de construction

14.2.1 Construire un triangle connaissant a, b et m la longueur de la bissectrice de l'angle des côtés a et b

Soit un triangle ABC et CM la bissectrice intérieure de l'angle C.
On pose $a = CB$, $b = CA$ et $m = CM$. Calculer en fonction de a et b.
On se donne trois longueurs a, b et m. On veut construire le triangle direct ABC dont m est la longueur de la bissectrice de l'angle des côtés a et b.
A quelle condition cela est-il possible ?
Lorsque cela est possible, faire cette construction avec `Xcas` comme si on utilisait le règle et le compas.
Une solution
On fait le dessin en tapant par exemple :

```
A:=point(4);
B:=point([1.536,1.865);
C:=point(0);
d:=bissectrice(C,A,B);
M:=inter_unique(droite(A;N),d)
```



On obtient :
On pose : $a = CB$, $b = CA$, $m = CM$, $x = AM$ et $y = BM$
Puisque CM est la bissectrice de l'angle C on a :
$\frac{x}{y} = \frac{b}{a}$ D'après le théorème d'Al Kashi, on a :
$x^2 = AM^2 = b^2 + m^2 - 2bm\cos(C/2)$ et
$y^2 = BM^2 = a^2 + m^2 - 2am\cos(C/2)$
Donc :
$\frac{x^2}{y^2} = \frac{b^2}{a^2} = \frac{b^2 + m^2 - 2bm\cos(C/2)}{a^2 + m^2 - 2am\cos(C/2)}$
Donc .
$b^2(a^2 + m^2 - 2am\cos(C/2)) = a^2(b^2 + m^2 - 2bm\cos(C/2))$
et puisque m n'est pas nul on en déduit :
$m(b^2 - a^2) = 2ab\cos(C/2)(b - a)$ ou encore
$m = \frac{2ab\cos(C/2)}{a + b}$ ou encore
$\cos(C/2) = \frac{m(a + b)}{2ab}$.
Puisqie l'angle $C/2$ est compris entre $-\pi/2$ et $\pi/2$, $\cos(C/2)$ est dans l'intervalle $]0, 1[$ donc la condition cherchée est :
$\frac{m(a + b)}{2ab} < 1$ ou encore $m < \frac{2ab}{(a + b)}$.
Comment faire la construcrtion du triangle ABC connaissant a, b et m ? Avec `Xcas`, il suffirait de définir l'angle C par : 2acos($\frac{m(a + b)}{2ab}$)
Mais on veut que cette construction se fasse comme avec la régle et le compas. On va donc mettre en évidence l'égalité :
$\frac{x}{y} = \frac{AM}{BM} = \frac{AC}{BC} = \frac{b}{a}$
Pour cela on méne par B la parallèle à CM, cette parallèle coupe AC en $B1$.
Puisque $\frac{AM}{BM} = \frac{AC}{BC}$ et que $\frac{AM}{BM} = \frac{AC}{CB1}$ on en déduit que :

$CB1 = CB = a$ et $BB1 = m\dfrac{a+b}{b}$.
La longueur $BB1$ est facile à construire avec la règle et le compas, la construction du triangle $CBB1$ est facile à construire avec la règle et le compas puisqu'on connait la longueur de ses 3 côtés. On en déduit ensuite le point A puisque $B1, C, A$ sont alignés et $CA = a$. D'où la construction du triangle ABC.
Avec `Xcas`, on tape :

```
supposons(a=[3.0,0,5,0.1]);
supposons(b=[4,0,5,0.1]);
supposons(m=[2.6,0,5,0.1]);
A:=point(a);
B1:=point(-b);
C:=point(0);
M1:=point(i*m);
D:=inter_unique(droite(A,M1),droite(B1,-b+i));
d:=normal(factor(longueur(B1,D)));
c1:=cercle(C,b):;c1;
c2:=cercle(B1,d):;c2;
B:=inter(c1,c2,M1);
triangle(A,B,C,affichage=1);
d:=bissectrice(C,A,B);
M:=inter_unique(d,droite(A,B),affichage=1);
normal(longueur(C,M));
```

On obtient :



On voit :
en noir la construction de la longueur $m\frac{a+b}{b}$ (on a $CM1 = m$ et $B1D = m\frac{a+b}{b}$.
en rouge le triangle ABC et M le pied de la bissectrice intérieure de l'angle C et `normal(longueur(C,M))` renvoie m.

14.3 Une transformation

Quelles sont les transformations du plan qui transforme toute droite en une droite parallèle ?
Ce qui veut dire que, si on connait un point `A` et son transformé `A1`, le transformé `B1` de `B`, est sur la parallèle à la `droite(A,B)` passant par `A1`. Si `A` et `A1` sont confondus en `O`, `B1` se trouve sur la `droite(O,B)` : `B`,`B1` et `O` sont alignes si `O` est un point fixe.
On va essayer de déterminer ces transformations en les classant selon le nombre de points fixes.
Soit `T` est une transformation du plan qui transforme toute droite en une droite parallèle et si,

— `T` a au moins deux points fixes `O1` et `O2`, alors le transformé d'un point `A` situé en dehors de la `droite(O1,O2)` est sur la `droite(A,O1)` et sur la `droite(A,O2)`, donc est en `A`. On en déduit que `A` est aussi un point fixe et que tous les points sont fixes puisque les points de la `droite(O2,O1)` sont en dehors de la droite la `droite(A,O1)` ou de la droite la `droite(A,O2)`.
) Donc si `T` a au moins deux points fixes, c'est que `T` est l'identité.

— `T` a un seul point fixe `O`, et soient deux points `A` et `B` non alignés avec `O`, et leur transformé `A1` et `B1`.
`A1` n'est pas confondu avec `O` car sinon `A1` et `B1` seraient confondus avec `O` car :
`B1:=inter_droite(droite(B,O),parallele(A1,droite(A,B)))`
et la `droite(A,B)` serait transformée en le `point(O)` ! `A1` est sur la `droite(A,O)` et `B1` est sur la `droite(B,O)`. On sait de plus que les `droite(A,B)` et `droite(A1,B1)` sont paralléles. Donc si `T` a au moins un seul point fixe, c'est que `T` est une homotétie.
Avec `xcas`, on clique pour définir deux points `A` et `B` et on tape :
```
O:=point(0);
t:=element(-2..5);
A1:=element(droite(A,O),t);
B1:=inter_droite(droite(B,O),parallele(A1,droite(A,B)));
```
puis on fait bouger `t` et `B`.

— `T` n'a pas de point fixe et soient deux points `A` et `B`, et leur transformé `A1` et `B1`.
La `droite(A,A1)` ne coupe pas la `droite(B,B1)` car sinon le point d'intersection `O` serait un point fixe : `O1:=inter_droite(droite(A,O),droite(B,O))` car `parallele(A1,droite(A,O))=droite(A,O)` et `parallele(B1,droite(B,O))=droite(B,O)`.
Donc :
`B1:=inter_droite(parallele(A1,droite(A,B)),parallele(B,droite(A,A1)))`
donc `ABA1B1` est un parallélogramme.
Donc si `T` n'a pas de point fixe, c'est que `T` est une translation.
Avec `Xcas`, on clique pour définir deux points `A` et `B` et on tape :
```
O:=point(0);
t:=element(-2..5);
A1:=element(droite(A,O),t);
```

```
B1:=inter_droite(parallele(A1,droite(A,B)),parallele(B,droite(A,
```

puis on fait bouger `t` et `B`.

14.4 L'inverseur de Peaucellier

Charles Peaucellier est un général français (1832-1913).
On appelle inverseur tout système articulé qui permet de tracer mécaniquement la figure inverse d'une figure plane donnée. L'inverseur de Peaucellier est constitué d'un losange articulé $AMBN$ de côté a. Aux sommets A et B sont articulés deux tiges OA et OB de longueur d avec $d \geq a$. Le point O est fixe et lorsque le point M se déplace sur un cercle passant par 0, le point N se déplace sur une droite qui est l'inverse du cercle dans l'inversion de centre O et de puissance $d^2 - a^2$.

Propriété :
Quand on fait bouger A et B, les points M et N restent alignés avec O et sont inverses l'un de l'autre dans l'inversion de centre O et de puissance $d^2 - a^2$.

14.4.1 Observation lorsque M décrit un cercle passant par O

On suppose que O est à l'origine, que $d = 3,\ a = 2$ et que M reste sur le cercle de centre 1 et de rayon 1.
On tape dans un écran de géométrie :

```
O:=point(0);
C:=cercle(0,point(2)):;C;
M:=element(C)
C1:=cercle(0,3.):;
C2:=cercle(M,2.):;
K:=inter(C1,C2):;
A:=K[0];B:=K[1];
segment(O,A,affichage=4);
segment(O,B,affichage=4);
N:=symetrie(droite(A,B),M);
L:=lieu(N,M);
quadrilatere(A,M,B,N,affichage=1);
```

On obtient :

On se met en mode `Pointeur` et on déplace M qui décrit le cercle. Le losange se déforme et N se déplace sur la droite inverse du cercle dans l'inversion de centre O et de puissance $\frac{5}{2}$

14.4.2 Démonstration

Les points M et N restent alignés avec O car O,M et N sont équidistants de A et B, ils sont donc sur la médiatrice de AB.
La puissance de O par rapport au cercle de centre A et de rayon $AM = AN = a$ est donc : $\overline{OM}.\overline{ON} = OA^2 - AM^2 = d^2 - a^2$.
Revenons à l'exemple précédent (O à l'origine, $d = 3, \ a = 2$). Le transformé du cercle de centre 1 et de rayon 1 par l'inversion de centre O et de puissance 5 ($3^2 - 2^2 = 5$) est une droite puisque ce cercle passe par O.
Cette droite a pour équation $x = \frac{5}{2}$ puisque le point du cercle d'abscisse 2 se transforme en le point de l'axe des x d'abscisse $\frac{5}{2}$.

14.5 Un pavage

14.5.1 Construction d'un pavage invariant par des translations

Soient 5 points `A,B,C,E,F`, On construit 3 points `D,G,H` par translation : `D` (resp `G`) est le transformé de `A` (resp `E`) dans la translation de vecteur `BC` et `H` est le transformé de `F` dans la translation de vecteur `BA`.
Le pavé de base est `P0=polygone([A,E,B,F,C,G,D,H])`.
Pour vous convaincre, on va exécuter le script suivant qui se trouve dans le fichier `pavage1.cxx` :

```
//un pave le polygone([A,E,B,F,C,G,D,H])
A:=point(-1.84,-1.83);
B:=point(0.22,-1.93);
C:=point(0,0);
E:=point(-1,-2);
F:=point(1.05,-0.857);
D:=translation(C-B,A);
G:=translation(C-B,E);
H:=translation(A-B,F);
nodisp(P0:=polygone(A,E,B,F,C,G,D,H));
nodisp(P1:=translation(B-A,P0));
P1;
translation(B-C,[P0,P1]);
```

vous pouvez faire bouger les points `A,B,C,E,F`

14.5.2 Avec un quadrilatère quelconque

Tout quadrilatère plan non croisé pave le plan.
Le pavé de base est `Q:=quadrilatere(A,B,C,D)`

Avec un script

Pour vous convaincre on va exécuter le script suivant qui dessine un quadrilatère quelconque `A,B,C,E` et ses représentants (son symétrique par rapport au milieu `O` de `AB` et ses translatés) formant un pavage.

```
//un quadrilatere quelconque pave le plan
A:=point(-1.84,-1.83);
B:=point(0.22,-1.93);
AB:=segment(A,B);
C:=point(1.05,-0.857);
BC:=segment(B,C);
D:=point(-0.0943,0.0178)+-0.0314-1.62*(i);
CD:=segment(C,D);
DA:=segment(D,A);
O:=milieu(A,B);
nodisp(Q:=quadrilatere(A,B,C,D));
nodisp(Q1:=symetrie(O,Q));
nodisp(Q2:=op(translation(D-B,[Q,Q1])));
Q;
Q1;
Q2;
translation(C-A,[Q,Q1,Q2]);
```

On met ce script comme commandes dans un niveau de géométrie (si vous avez tapé ce script est dans un èditeur de programme, en ayant pris soin de n'écrire

qu'une seule commande (terminée par `;`) par ligne, vous pouvez mettre ces commandes d'un coup de souris dans les lignes de commandes dans un niveau de géométrie : on sélectionne le script et on clique sur le numero d'une ligne de commandes dans un niveau de géométrie et cela recopie le script à partir de cette ligne).
Puis on coche le bouton `Step` pour exécuter le script pas à pas.
Vous pouvez déformer ce quadrilatère en faisant bouger l'un des points `A,B,C,E`.

Sur le même principe, on peut réaliser un pavage en remplacant les côtés du quadrilatère par des lignes brisées admettant un centre de symétrie.
Pour vous convaincre on va exécuter le script suivant :

```
//un "quadrilatere" chaque cote est invariant par symetrie centrale
A:=point(-1.84,-1.83);
B:=point(0.22,-1.93);
C:=point(1.05,-0.857);
D:=point(-0.0943,0.0178)
M:=milieu(A,B);
N:=milieu(C,B);
O:=milieu(C,D);
P:=milieu(A,D);
E:=point(-1.2,-2);
F:=point(0.6,-1.8);
G:=point(0.8,-0.5);
H:=point(-0.5,0);
nodisp(E1:=symetrie(M,E));
nodisp(F1:=symetrie(N,F));
nodisp(G1:=symetrie(O,G));
nodisp(H1:=symetrie(P,H));
nodisp(P0:=polygone(A,E,M,E1,B,F,N,F1,C,G,O,G1,D,H,P,H1,A));
P0;
translation(A-C,P0);
```

vous pouvez faire bouger les points `A,B,C,D,E,F,G,H`.

14.5.3 Avec une animation

On peut aussi écrire le programme `pavage` dans un éditeur de programmes qui va réaliser à partir du quadrilatère A, B, C, D, un pavage de l lignes et de c colonnes :

```
pavage(A,B,C,D,l,c):={
local k,LP,LQ,LLP,LLQ,,P,Q;
P:=polygone(A,B,C,D);
Q:=symetrie(milieu(A,B),P);
LP:=P;
LQ:=Q;
pour k de 1 jusque c-1 faire
P:=translation(C-A,P);
```

```
Q:=translation(C-A,Q);
LP:=LP,P;
LQ:=LQ,Q;
fpour;
LLP:=LP;
LLQ:=LQ;
LP:=[LP];
LQ:=[LQ];
pour k de 1 jusque l-1 faire
LP:=translation(B-D,LP);
LLP:=LLP,op(LP);
LQ:=translation(B-D,LQ);
LLQ:=LLQ,op(LQ);
fpour;
return [affichage(LLP,1+rempli),affichage(LLQ,2+rempli)];
}:;
```

On compile ce programme avec `F9` et on ouvre un écran de géométrie : On se met en mode `point` et on clique pour obtenir 4 points `A,B,C,D`. On tape ensuite
`pavage( A,B,C,D,6,8)`
On peut enlever les légendes `A,B,C,D` et les axes (bouton `cfg`, puis décocher `Montrer les noms` et décocher `Montrer les axes`).
On se met en mode `pointeur` et on déplace un des points `A,B,C,D`. ou bien on rajoute un paramètre `t` et on tape par exemple : `t:=element(0 .. 6.3,1.8,0.1)`
`pavage(0,1,0.5+i*0.5,0.25+i*0.75+0.25*sqrt(2)*exp(i*t),5,5)`
et on fait bouger le curseur `t`
ou encore on fait une animation et on tape :
`L:=seq(pavage(0,1,0.5+i*0.5,0.25+i*0.75+0.25*sqrt(2)*exp(i*t),5,5),t=0`
`animation(L);`
Le temps entre 2 images est défini dans `cfg->animate`.

On peut aussi ouvrir un niveau de géométrie 2-d et écrire les commandes :

```
A:=point(0);
B:=point(1);
C:=point(0.5+i*0.5);
t:=element(0 .. 12.6,12.6,0.1);
D:=point(,0.25+i*0.75+0.25*sqrt(2)*exp(i*t));
pavage(A,B,C,D,5,8)
```

Puis on fait bouger `t` et le pavage se déforme. Pour que cette déformation se fasse automatiquement, on utilise le menu `M->Animation->Gaph off` pour enlever le dessin, puis `M->Animation->Creer animation` pour créer l'animation (on vous demande combien vous voulez d'images (`frame`) différentes), la vitesse de défilement est définie par `cfg->animate`.

14.5.4 Construction d'un pavage triangulaire

Le pavé de base est `P0=polygone([A,E,B,F,C,H,J,G])` Pour vous convaincre on va exécuter le script suivant qui se trouve dans le fichier `pavage3.cxx` :

```
//un pave P0= polygone([A,E,B,F,C,H,J,G])([A,E,B,F,C,H,J,G])
A:=point(-1.84,-1.83);
B:=point(0.22,-1.93);
nodisp(triangle_equilateral(A,B,C));
E:=point(-1,-2);
F:=point(0,0);
G:=rotation(A,2*pi/3,E);
H:=rotation(C,-2*pi/3,F);
J:=rotation(C,-2*pi/3,B);
nodisp(P:=[A,E,B,F,C,H,J,G]);
nodisp(P0:=polygone(op(P)));
nodisp(P1:=rotation(A,2*pi/3,P0));
nodisp(P2:=rotation(A,4*pi/3,P0));
[P0,P1,P2];
translation(B-J,[P0,P1,P2]);
translation(B-rotation(A,2*pi/3,J),[P0,P1,P2]);
```

vous pouvez faire bouger les points `A,B,E,F`

14.5.5 Construction d'un pavage carre

Le pavé de base est `P0=polygone([A,E,B,F,C,H,J,G])` Pour vous convaincre on va exécuter le script suivant qui se trouve dans le fichier `pavage4.cxx` :

```
//un pave P0=polygone([A,E,B,F,C,H,J,G])
A:=point(-1.84,-1.83);
B:=point(0.22,-1.93);
nodisp(C:=similitude(A,sqrt(2)/2,pi/4,B));
E:=point(-1,-2);
F:=point(0,-1.2);
G:=rotation(A,pi/2,E);
H:=rotation(C,-pi,F);
J:=rotation(C,-pi,B);
nodisp(P:=[A,E,B,F,C,H,J,G]);
nodisp(P0:=polygone(op(P)));
nodisp(P1:=rotation(A,pi/2,P0));
nodisp(P2:=rotation(A,pi,P0));
nodisp(P3:=rotation(A,3*pi/2,P0));
[P0,P1,P2,P3];
translation(B-J,[P0,P1,P2,P3]);
translation(2*(B-A),[P0,P1,P2,P3]);
translation(B-rotation(A,pi,J),[P0,P1,P2,P3]);
```

vous pouvez faire bouger les points `A,B,E,F`

14.6 Le pentagone, $\sin(\frac{\pi}{5})$, $\sin(\frac{2\pi}{5})$, $\cos(\frac{\pi}{5}$ et $\cos(\frac{2\pi}{5})$

14.6.1 Calcul exact de $\sin(2\pi/5)$ et de $\cos(2\pi/5)$

Si $a = \exp(2i\pi/5) = \cos(2\pi/5) + i\sin(2\pi/5)$,on a :
$a^5 - 1 = (a-1)(a^4 + a^3 + a^2 + a + 1) = 0$.
Comme $a \neq 1$, on a a vérifie :
$a^4+a^3+a^2+a+1 = a^2(a^2+1/a^2+a+1/a) = a^2((a+1/a)^2+(a+1/a)-1) = 0$
Comme $a \neq 0$, on a $2\cos(2\pi/5) = a + 1/a$ vérifie l'équation :
$z^2 + z - 1 = 0$
On tape :
```
solve(z^2+z-1=0,z)
```
On obtient :
```
[1/2*(-1-(sqrt(5))),1/2*(-1+sqrt(5))]
```
Comme $2\cos(2\pi/5) > 0$ et $\sin(2\pi/5) = \sqrt{1-\cos(2\pi/5)^2}$ on tape :
```
normal(1-(1/4*(-1+sqrt(5)))^2)
```
On obtient :
```
(sqrt(5)+5)/8
```
et on en déduit que :

$$\cos(2\pi/5) = \frac{\sqrt{5}-1}{4}$$

$$\sin(2\pi/5) = \sqrt{\frac{\sqrt{5}+5}{8}}$$

On aurait aussi pu taper en mode complexe :
```
solve(a+1/a=1/2*(-1+sqrt(5)),a)
```
On obtient comme valeur de a :
```
[1/4*(sqrt(5)-1+sqrt(2*sqrt(5)+10)*(i)),
1/4*(sqrt(5)-1-sqrt(2*sqrt(5)+10)*(i))]
```
Comme $a = \exp(2i\pi/5) = \cos(2\pi/5) + i\sin(2\pi/5)$, on a $re(a > 0$ donc :

$$a = \cos(2\pi/5) + i\sin(2\pi/5) = \frac{\sqrt(5) - 1 - i\sqrt{2\sqrt(5) + 10}}{4}$$

14.6.2 Calcul exact de $\sin(\pi/5)$ et de $\cos(\pi/5)$

On déduit de ce qui précède les valeurs de $\cos(\pi/5)$ et de $\sin(\pi/5)$ puisque :
$\cos(2\pi/5) = 2\cos(\pi/5)^2 - 1$ et $\cos(2\pi/5) = 1 - 2\sin(\pi/5)^2$.
On tape :
```
normal(sqrt((sqrt(5)-1)/8+1/2))
```
On obtient la valeur de $\cos(\pi/5)$:
```
(sqrt(5)+1)/4
```
On tape :
```
normal(-(sqrt(5)-1)/8+1/2)
```
On obtient $\sin(\pi/5)^2$:
```
(-(sqrt(5))+5)/8
```
Donc :

$$\cos(\pi/5) = \frac{\sqrt{5}+1}{4}$$

$$\sin(\pi/5) = \sqrt{\frac{-\sqrt{5}+5}{8}}$$

14.6.3 Construction du pentagone comme avec la règle et le compas

On tape :

```
R:=1;
C:=cercle(0,R);
A:=point(-R);
B:=point(i*R);
m:=milieu(0,B);
c1:=cercle(m,R/2,affichage=2);
segment(A,m,affichage=2);
I:=inter(c1,segment(m,A));
r:=longueur2(A,I);
c:=cercle(A,sqrt(r));
L:==simplify(inter(C,c)):;
P:=L[0];
Q:=L[1];
segment(P,Q);
c2:=cercle(P,longueur(P,Q)):;
affichage(c2,4);
M:=inter(C,c2,B);
c3:=cercle(Q,longueur(P,Q)):;
affichage(c3,4);
T:=point(R);
S:=inter(C,c3,T);
polygone(T,M,P,Q,S,affichage=1);
```

On obtient :



On trouve aussi que `longueur2(A,I)` vaut :
`(-1-(-4*sqrt(5))/20)^2+(-(-2*sqrt(5)+10)/20)^2`
On tape :
`simplify(re(affixe(P)));`
On obtient :
`(-(sqrt(5))-1)/4`
c'est la vaeur de $-\cos(\pi/5)$
On tape :
`simplify(im(affixe(P))^2);`
On obtient :
`(-(sqrt(5))+5)/8`
c'est la vaeur de $\sin(\pi/5)^2$
P a donc comme coordonnèes : $(-\cos(\pi/5); \sin(\pi/5))$
Donc le polygone T, M, P, Q, S est bien un pentagone régulier.

14.7 Des étoiles à 5 branches

14.7.1 Une étoile à 5 branches

On cherche tout d'abord la liste des sommets du polygone étoile à 5 branches : les pointes (resp les creux) se déduisent par rotation d'angle $2 * \pi/5$. On définit ainsi les sommets d'un polygone puis, on affiche ce polygone avec le programme `etoil`. Si on remplit le polygone `etoil`, il devient le polygone `etoile`.
On va utiliser 3 paramètres :
`z0` le centre de l'étoile,
`r` le rayon de l'étoile,
`a` l'argument d'un "sommet en creux" de l'étoile,

Ces paramètres permettent de positionner l'étoile dans le plan. On calcule la distance `l` d'un "sommet en creux" au centre de l'étoile :
on sait ou on retrouve (puisque $1 + 2 * \cos(2 * \pi/5) + 2 * \cos(4 * \pi/5) = 0$) que :
$\cos(2 * \pi/5) = (\sqrt{5} - 1)/4$
$\cos(\pi/5)^2 = (3 + \sqrt{5})/8$
$\cos(\pi/5) = (1 + \sqrt{5})/4$
on a :
$l = \cos(2 * \pi/5)/\cos(\pi/5)$
donc on a :
$l := r(3 - \sqrt{(}5))/2$
On a : $(\sqrt{5} - 1)/2 \sim 0.61803398875$



On tape :

```
etoil(z0,r,a):={
  local j,l,somet,p,L,pa;
  z0:=evalf(z0);r:=evalf(r);a:=evalf(a);
  l:=evalf(r*(3-sqrt(5))/2);
  somet:=[z0+l*exp(i*a),z0+r*exp(i*(a+evalf(pi)/5))];
  L:=somet;
  for (j:=1;j<5;j++){
    L:=concat(L,rotation(z0,2*j*evalf(pi)/5,somet));
  }
  p:=polygone(L);
  return p;
}:;

etoile(z0,r,a):={
  return affichage(etoil(z0,r,a),rempli);
}:;
```

On tape :
`etoile(0,1,0)`
`etoile(3,2,pi/5)`
On obtient :



14.7.2 Une étoile faite d'étoiles

On veut faire le dessin :



Si le rayon de l'étoile centrale est R et celui de l'étoile suivante est r on a la relation : $r * \sin(2 * \pi/5) = R * \sin(\pi/5)$ ($R \sim 1.61803398875 * r$)
et on a trouvé que les sommets en "creux" sont situés sur un cercle de rayon : $l = R * (3 - \sqrt{5})/2$
On sait que : $\sin(2 * \pi/5)^2 = 1 - \cos(2 * \pi/5)^2 = (5 + \sqrt{5})/8$ et
$\sin(\pi/5)^2 = 1 - \cos(\pi/5)^2 = (5 - \sqrt{5})/8$
donc $(5 + \sqrt{5}) * r^2 = R^2 * (5 - \sqrt{5}) = 20/(5 + \sqrt{5})$ soit
$r = 2 * R * \sqrt{5}/(5 + \sqrt{5}) = 2 * R/(1 + \sqrt{5}) = R * (\sqrt{5} - 1)/2$
De plus le centre de l'étoile suivante est situé à :
$l + r = R * (3 - \sqrt{5})/2 + R * (-1 + \sqrt{5})/2 = R.$
On tape :

```
etoiles(z0,r,a):={
  local j,k,R,L,nr,nz0;
  L:=[etoile(z0,r,a)];
  R:=r;
  for (j:=0;j<5;j++){
```

```
    nr:=2*R/(1+sqrt(5));
    nz0:=z0+R*exp(2*i*j*pi/5+i*a);
    for (k:=1;k<5;k++){
      L:=append(L,etoile(evalf(nz0),nr,a));
      r:=nr;
      nr:=2*r/(1+sqrt(5));
      nz0:=nz0+r*exp(2*i*j*pi/5+i*a);
    }
  }
  return L;
}:;
```

On tape :

`etoiles(0,1,0);`

`affichage(etoil(0,2/(3-sqrt(5)),pi/5),line_width_3)`

On obtient le dessin voulu.

On tape :

`etoiles(0,1,pi/4);`

`affichage(etoil(0,2/(3-sqrt(5)),pi/5+pi/4),line_width_3)`

On obtient le même dessin tourné de $\pi/4$.

Si on veut faire la même chose avec une ètoile à 7 branches on tape :

```
etoil7(z0,r,a):={
  local j,l,somet,p,L,pa;
  z0:=evalf(z0);
  r:=evalf(r);
  a:=evalf(a);
  //l:=evalf(r*(3-sqrt(5))/2);
l:=evalf(r*cos(2*pi/7)/cos(pi/7));
  somet:=[z0+l*exp(i*a),z0+r*exp(i*(a+evalf(pi)/7))];
  L:=somet;
  for (j:=1;j<7;j++){
    L:=concat(L,rotation(z0,2*j*evalf(pi)/7,somet));
  }
  p:=polygone(L);
  return p;
}:;

etoile7(z0,r,a):={
  return affichage(etoil7(z0,r,a),rempli);
}:;
etoiles7(z0,r,a):={
  local j,k,R,L,nr,nz0,nl,l;
  L:=[etoile7(z0,r,a)];
  R:=r;
  l:=evalf(R*cos(2*pi/7)/cos(pi/7));
for (j:=0;j<7;j++){
    nr:=evalf(R*sin(pi/7)/sin(2*pi/7));
```

```
    nz0:=z0+(l+nr)*exp(2*i*j*pi/7+i*a);
    for (k:=1;k<7;k++){
      L:=append(L,etoile7(evalf(nz0),nr,a));
      r:=nr;
      nr:=r*sin(pi/7)/sin(2*pi/7);
nl:=evalf(r*cos(2*pi/7)/cos(pi/7));
      nz0:=nz0+(nl+nr)*exp(2*i*j*pi/7+i*a);
    }
  }
  return L;
}:;
```

14.7.3 Le logo de Xcas

Le logo de Xcas est obtenu en tapant :

```
etoilo(z0,r,a):={
  local j,l,somet,p,L,pa;
  z0:=evalf(z0);
  r:=evalf(r);
  a:=evalf(a);
  l:=evalf(r*(3-sqrt(7))/2);
  somet:=[z0+l*exp(i*a),z0+r*exp(i*(a+evalf(pi/7)))];
  L:=somet;
  for (j:=1;j<7;j++){
    L:=concat(L,rotation(z0,2*j*evalf(pi/7),somet));
  }
  p:=polygone(L);
  return p;
}:;
etoilog(z0,r,a):={
  return affichage(etoilo(z0,r,a),rempli);
}:;
logox(z0,r,a,c):={
  local j,k,R,L,nr,nz0;
  L:=[affichage(etoilo(z0,r,a),c+rempli)];
  R:=r;
  for (j:=0;j<7;j++){
    nr:=2*R/(1+sqrt(7));
    nz0:=z0+R*exp(2*i*j*pi/7+i*a);
    for (k:=1;k<7;k++){
      L:=append(L,affichage(etoilo(evalf(nz0),nr,a),
                            c+(j+1)*k+rempli));
      r:=nr;
      nr:=2*r/(1+sqrt(7));
      nz0:=nz0+r*exp(2*i*j*pi/7+i*a);
    }
  }
  return L;
```

```
}:;
 lx(z0,r):={
  return(segment(z0+r*(-1-i),z0+r*(1+i)),
         segment(z0+r*(1-i),z0+r*(-1+i)));
}:;
lc(z0,r):={
  return (cercle(z0,r,pi/4,7*pi/4));
}:;
la(z0,r):={
  return(segment(z0+r*(-1-i),z0+r*i),
         segment(z0+r*(1-i),z0+r*i),
         segment(z0+r*-0.5,z0+r*0.5));
}:;
ls(z0,r):={
  return (segment(z0+r*(-1/2-i),z0-r*i),
          segment(z0+r*(1/2+i),z0+r*i),
          cercle(z0+r*i/2,r/2,pi/2,3*pi/2),
          cercle(z0-r*i/2,r/2,-pi/2,pi/2));
}:;
logoxcas(z0,r,a,c):={
return logox(z0,r,a,c),
affichage(lx(evalf(z0-2*r*exp(i*a),r*0.2)),
           line_width_3+c+4),
affichage(lc(evalf(z0-2*r*exp(-2*i*pi/7+i*a),0.2*r)),
             line_width_3+c+3),
affichage(la(evalf(z0-2*r*exp(-4*i*pi/7+i*a),0.2*r)),
             line_width_3+c+2),
affichage(ls(evalf(z0-2*r*exp(-6*i*pi/7+i*a),0.2*r)),
             line_width_3+c+1);
}:;
```

On tape :
`logoxcas(0,1,0,264);`
On obtient les 7 branches de `Xcas` :

- Calcul formel
- Tableur formel
- Géométrie 2D interactive
- Géométrie 3D interactive
- Géométrie Tortue
- Langage de programmation
- Documentation



14.7.4 La carte de visite de `Xcas`

Voici ce qu'il faut taper pour avoir la carte de visite de `Xcas` :

```
logoxcas(0,0.9,pi/14,264);
legende(-2.2+3i,": le couteau suisse des mathematiques");
legende(-4.5+0.4i,"programmation",magenta);
legende(1.2+1.8i,"geometrie 2d",cyan);
legende(-4-1.2i," tableur formel",rouge);
legende(-3+1.8i,"calcul formel",bleu);
legende(2.3+0.4i,"geometrie 3d",jaune);
legende(1.7-1.2i,"geometrie tortue",vert);
legende(-1.1-2.1i,"documentation");
legende(-4.6+2.5i,"http://www-fourier.ujf-grenoble.fr
                          /~parisse/giac_fr.html");
rectangle(-5-2.2i,5-2.2i,5/8.5);
affichage([lx(-4+3.2i,0.2),lc(-3.5+3.1i,0.2),la(-3+3.2i,0.2),
           ls(-2.5+3.1i,0.2)],epaisseur_ligne_3);
```

On obtient :



14.7.5 La carte de Noel de `Xcas`

Voici ce qu'il faut taper pour avoir la carte de Noel de `Xcas` :

```
lx(z0,r):={
  return(segment(z0+r*(-1-i),z0+r*(1+i)),
```

```
         segment(z0+r*(1-i),z0+r*(-1+i)));
}:;
lc(z0,r):={
  return (cercle(z0,r,pi/4,7*pi/4));
}:;
la(z0,r):={
  return(segment(z0+r*(-1-i),z0+r*i),
         segment(z0+r*(1-i),z0+r*i),
         segment(z0+r*-0.5,z0+r*0.5));
}:;
ls(z0,r):={
  return (segment(z0+r*(-0.6-i),z0+r*0.02-r*i),
          segment(z0+r*(0.6+i),z0+r*i),
          cercle(z0+r*i/2,r*1.02/2,pi/2,3*pi/2),
          cercle(z0-r*1.01*i/2,r*1.02/2,-pi/2,pi/2));
}:;
ls1(z0,r):={
  return (segment(z0+r*(-1-i),z0+r*0.55-r*i),
          segment(z0+r*(1+i),z0-0.5*r+r*i),
           segment(z0-0.55*r,z0+0.55*r),
          cercle(z0-r*0.5+r*i*0.5,r*0.51,pi/2,3*pi/2),
          cercle(z0+r*0.5-r*i*0.5,r*0.52,-pi/2,pi/2));
}:;
lm(z0,r):={
  return polygone_ouvert(z0,z0+i*r,z0+(1+i)/2*r,z0+(1+i)*r,z0+r);
}:;
le(z0,r):={
  return polygone_ouvert(z0+3*r/4,z0,z0+i*r,z0+r*i+r*3/4),
         segment(z0+r*i/2,z0+r*i/2+3*r/4);
}:;
ly(z0,r):={
  return polygone_ouvert(z0+r*i,z0+(1+i)*r/2,z0+r/2,
                         z0+(1+i)*r/2,z0+(1+i)*r);
  }:;
lr(z0,r):={
  return cercle(z0+(0.5+i*0.75)*r,r*0.25,-pi/2,pi/2),
  polygone_ouvert(z0+r/2+i*r,z0+i*r,z0,z0+i*r/2,
                  z0+r/2+i*r/2,z0+r*3/4);
}:;
etoilo(z0,r,a):={
  local j,l,somet,p,L,pa;
  z0:=evalf(z0);r:=evalf(r);a:=evalf(a);
  l:=evalf(r*(3-sqrt(7))/2);
  somet:=[z0+l*exp(i*a),z0+r*exp(i*(a+evalf(pi/7)))];
  L:=somet;
  for (j:=1;j<7;j++){
    L:=concat(L,rotation(z0,2*j*evalf(pi/7),somet));
  };
```

```
  p:=polygone(L);
  return p;
}:;
etoilog(z0,r,a):={
  return affichage(etoilo(z0,r,a),rempli);
}:;
logox(z0,r,a,c):={
  local j,k,R,L,nr,nz0;
  L:=[affichage(etoilo(z0,r,a),c+rempli)];
  R:=r;
  for (j:=0;j<7;j++){
    nr:=2*R/(1+sqrt(7));
    nz0:=z0+R*exp(2*i*j*pi/7+i*a);
    for (k:=1;k<7;k++){
      L:=append(L,affichage(etoilo(evalf(nz0),nr,a),
                            c+(j+1)*k+rempli));
      r:=nr;
      nr:=2*r/(1+sqrt(7));
      nz0:=nz0+r*exp(2*i*j*pi/7+i*a);
    }
  }
  return L;
}:;
logoxcas(z0,r,a,c):={
  return logox(z0,r,a,c),
         affichage(lx(evalf(z0-2*r*exp(i*a),r*0.2)),
           line_width_3+c+4),
          affichage(lc(evalf(z0-2*r*exp(-2*i*pi/7+i*a),0.2*r)),
             line_width_3+c+3),
          affichage(la(evalf(z0-2*r*exp(-4*i*pi/7+i*a),0.2*r)),
             line_width_3+c+2),
          affichage(ls(evalf(z0-2*r*exp(-6*i*pi/7+i*a),0.2*r)),
             line_width_3+c+1);
}:;
cartev(z0,r):={
  local L;
  L:=lm(z0+1+4*i,r),le(z0+1+3*r/2+4*i,r),lr(z0+1+11*r/4+4*i,r),
     lr(z0+1+4*r+4*i,r),ly(z0+1+21*r/4+4*i,r);
  L:=L,lx(z0+1+2*i,r),lc(z0+1+2r+r/2+2*i,r),
      la(z0+1+4*r+r/2+2*i,r),ls(z0+1+13*r/2+2*i,r);
    return L;
  }:;
support(z0,r):={
  return segment(z0+r*(0.9+i*0.45),z0-r*(1.2+i*0.6)),
            segment(z0,z0-2*r*i);
  }:;
bulle(z0,r):={
  return affichage(support(z0,r),264+epaisseur_ligne_3),
```

```
          logoxcas(z0-3*r*i,0.42*r,0,264),cercle(z0-3*r*i,r);
  }:;
cartev1(c1,c2):={
local L;
L:=affichage(cartev(-2,1),59+epaisseur_ligne_4),
            rectangle(-4-i,8-i,0.67);
L:=L,logox(j,0.3,-pi/7,c1)$(j=-3..7);
L:=L,logox(j+6*i,0.3,pi/7,c1)$(j=-3..7);
L:=L,logox(-3+j*i,0.3,pi/7,c2)$(j=1..5);
L:=L,logox(7+j*i,0.3,pi/7,c2)$(j=1..5);
return L;
}:;
cartev2():={
  local L;
  L:=rectangle(-4.25-0.5*i,8-0.5*i,0.67),bulle(-0.5+4*i,1);
  L:=L,bulle(1.5+5*i,1),bulle(3.5+6*i,1),bulle(5.5+7*i,1);
  L:=L,affichage([lx(-2.5+5*i,0.5),lc(-2.5+3.5*i,0.5),
                 la(-2.5+2*i,0.5),ls1(-2.5+0.5*i,0.5)],
                 264+epaisseur_ligne_4);
  L:=L,affichage([lm(-3.75+6.3*i,0.75),le(-2.25+6.3*i,0.75),
                lr(-1+6.3*i,0.75),lr(0.25+6.3*i,0.75),
                ly(1.5+6.3*i,0.75)],232+epaisseur_ligne_4);
return L;
}:;
sapin(z0,z1,t):={
  local L,v;
    L:=NULL;v:=z1-z0;
    si abs(v)<0.2 alors L:=L, segment(z0,z1);retourne L; fsi;
    L:=L,sapin(z0+v/4.,z1,t);
    L:=L,segment(z0,z0+v*0.25);
    L:=L,sapin(z0,z0+v*exp(i*t)*0.5,t);
    L:=L,sapin(z0,z0+v*exp(-i*t)*0.5,t);
    }:;
cartev3():={
 retourne  affichage([lx(-4.5+5*i,0.5),lc(-4.5+3.5*i,0.5),
 la(-4.5+2*i,0.5),ls1(-4.5+0.5*i,0.5)],264+epaisseur_ligne_4),
affichage([lm(-5.75+6.3*i,0.75),le(-4.37+6.3*i,0.75),
lr(-3.06+6.3*i,0.75),lr(-1.72+6.3*i,0.75),
ly(-0.375+6.3*i,0.75)],232+epaisseur_ligne_4),
affichage(sapin(0,6.5*i,1),60),rectangle(-6.25-i,4-i,0.86),
logoxcas(0+6.5*i,0.45,0,269),
affichage(etoilog(-1.9+2.9*i,0.2,0),1),
affichage(etoilog(1.9+2.9*i,0.2,0),3),
affichage(etoilog(2.5+1.6*i,0.2,0),5),
affichage(etoilog(0.5+1*i,0.2,0),1),
affichage(etoilog(-1+0.5*i,0.2,0),3),
affichage(etoilog(-2.5+i*1.6,0.2,0),6);
}:;
```

Puis on tape :
`cartev1(86,88)`
On obtient une carte de Noel.
Puis on tape : `cartev2(),logoxcas(-2.5+3.5*i,0.5,0,264)`
On obtient une carte de Noel. Puis on tape : `cartev3()`
On obtient une carte de Noel.

14.7.6 Encore des étoiles à 5 branches

On veut réaliser :



Pour cela, on va utiliser le programme précédent :

```
//z0 centre, r rayon de l'etoile,
//a argument d'un "sommet en creux" de l'etoile
etoil(z0,r,a):={
  local j,l,somet,p,L,pa;
  z0:=evalf(z0);r:=evalf(r);a:=evalf(a);
  l:=evalf(r*(3-sqrt(5))/2);
  somet:=[z0+l*exp(i*a),z0+r*exp(i*(a+evalf(pi)/5))];
  L:=somet;
  for (j:=1;j<5;j++){
    L:=concat(L,rotation(z0,2*j*evalf(pi)/5,somet));
  }
  p:=polygone(L);
  return p;
}:;
etoile(z0,r,a):={
```

```
  return affichage(etoil(z0,r,a),rempli);
}:;
```

et les calculs précédents et écrire la procédure `etoil10` de paramètre `c` le centre de l'étoile pour dessiner :



On pose :
$a := 2 * \cos(\pi/5) = \sin(2 * \pi/5)/\sin(\pi/5) \simeq 1.61803398875$
et on a :
$R = a * r$ si r, R sont les rayons des cercles circonscrits à 2 étoiles consécutives,
Il faut maintenant trouver la relation entre le rayon r d'une ètoile et le rayon ρ du cercle sur lequel on va placer le centre de 10 étoiles.
Soit la figure ci-dessous :
il y a un pentagone (il définit une étoile à 5 branches) de côté AB et de rayon $r = OA$. Si on peut mettre 10 étoiles de rayon r sur le cercle de rayon $\rho = IO$ c'est que l'angle $\widehat{AIB} = \pi/10$:



On cherche la relation qui existe entre $IO = \rho$ et $r = OA$.
On a : $AB = 2 * r * \sin(\pi/5)$ et $IA = \rho + r$ donc :
$\dfrac{AB}{\sin(\pi/10)} = \dfrac{IA}{\sin(3\pi/5)}$ donc :
$2 * r * \sin(\pi/5) * \sin(3\pi/5)/\sin(\pi/10) = \rho + r$
donc :
$r = cr * \rho$ avec $cr = \dfrac{\sin(\pi/10)}{2\sin(\pi/5) * \sin(3\pi/5) - \sin(\pi/10)} \simeq 0.38196601125$
On tape :

```
etoil5(c):={
local a,cr;
a:=evalf(2*cos(pi/5));
cr:=sin(pi/10.)/(2*sin(pi/5.)*sin(3*pi/5.)-sin(pi/10.));
return (etoil(c+exp(i*k*pi/5),cr,(k+1)*pi/5))$(k=0..9),
(etoil(c+a*exp(i*k*pi/5),cr*a,(k+1)*pi/5))$(k=0..9),
(etoil(c+a^2*exp(i*k*pi/5),cr*a^2,(k+1)*pi/5))$(k=0..9),
(etoil(c+1/a*exp(i*k*pi/5),cr/a,(k+1)*pi/5))$(k=0..9);
```

On tape :

```
a:=evalf(2*cos(pi/5));b:=a^3
cr:=sin(pi/10.)/(2*sin(pi/5.)*sin(3*pi/5.)-sin(pi/10.));
etoil5(b*exp(i*k*2*pi/5))
```

On obtient :

```
etoil5(b*exp(i*k*2*pi/5))$(k=0..4),etoil(0,b*cr,0)
```

On obtient :



On peut ainsi faire une sorte de pavage mais on est obligé de ruser pour qu'il n'y ait pas de chevauchement :
`etoil10` on enleve la dernière couronne à `etoil5`.
`etoils5` est formé de 5 `etoil5`
`etoils15` est formé de 1 `etoil5` et de 4 `etoil10` et `etoil5` est à la $n$ième position.
On tape :

```
etoil10(c):={
local cr,a:=evalf(2*cos(pi/5));
cr:=sin(pi/10.)/(2*sin(pi/5.)*sin(3*pi/5.)-sin(pi/10.));
return (etoil(c+exp(i*k*pi/5),cr,(k+1)*pi/5))$(k=0..9),
(etoil(c+a*exp(i*k*pi/5),cr*a,(k+1)*pi/5))$(k=0..9),
(etoil(c+1/a*exp(i*k*pi/5),cr/a,(k+1)*pi/5))$(k=0..9);
}
:;
etoils5(c,t):={
local a,b,cr;
a:=evalf(2*cos(pi/5));
b:=a^3;
cr:=sin(pi/10.)/(2*sin(pi/5.)*sin(3*pi/5.)-sin(pi/10.));
return etoil5(c+b*exp(i*k*2*pi/5+i*t))$(k=0..4),
        etoil(c,b*cr,t);
}:;
etoils15(c,t,n):={
```

```
local a,b,cr;
a:=evalf(2*cos(pi/5));
b:=a^3;
cr:=sin(pi/10.)/(2*sin(pi/5.)*sin(3*pi/5.)-sin(pi/10.));
return etoil10(c+b*exp(i*k*2*pi/5+i*t))$(k=n+1..n+4),
       etoil(c,b*cr,t),etoil5(c+b*exp(i*n*2*pi/5+i*t));
}:;
```

On tape :
`etoils5(0,0),(etoils15(2*b*sin(3*pi/10)*exp(i*pi/5+2*i*k*pi/5),pi/5,k`
On obtient :



14.7.7 Le logo de `l'UJF`

```
arcpoly(z0,r,a,b):={
local L;
return seq(z0+r*exp(i*t),t=a..b,0.05),z0+r*exp(i*b);
}:;
arc_poly(z0,r,a,b,ep,c):={
local L;
L:=z0+(r-ep)*exp(i*a),z0+r*exp(i*a),arcpoly(z0,r,a,b),
  z0+(r)*exp(i*b),z0+(r-ep)*exp(i*b),arcpoly(z0,r-ep,b,a);
return affichage(polygone(L),c+rempli);
}:;
ujf(z0,r):={
local L;
```

```
L:=carre(z0,z0+r,affichage=48+rempli),carre(z0+r/3*(-1+2*i),z0+2*r*i/3,affich
L:=L,rectangle(z0+r/3*(-2+i),z0+r*i/3,0.5,affichage=1+rempli),rectangle(z0-r*
L:=L,rectangle(z0+-r/3*(1+i),z0-r*i/3,3,affichage=1+rempli),arc_poly(z0-r,r,-
L:=L,polygone(z0-r/6-r*i,z0+r/6-r*i,z0+r/3,z0,affichage=1+rempli);
L:=L,affichage(polygone(z0+r*(0.9+i),arcpoly(z0+17*r/12,17*r/12,pi-atan(5/5),
arcpoly(z0+17*r/12,13*r/12,pi,pi-atan(12/5))),1+rempli);
L:=L,legende(z0+r+i*2r/3,"UNIVERSITE"),legende(z0+r+0.1*i,"JOSEPH FOURIER"),l
legende(z0+r+i*0.1," ________________",affichage=1);
return(L);
}:;
```

On tape :
`ujf(0,1)`
On obtient :



14.8 Un quasi cristal

Sur le pourtour d'un décagone on trace des pentagones, puis des étoiles à 5 branches, puis un tour de décagones avec ses pentagones puis des étoiles et un tour de décagones avec ses pentagones.
La fonction `etoile` a comme pramètres :
`a` est l'affixe de son centre,
`r` est le rayon de son cercle circonscrit,
`k` est l'angle en radian que fait une branche de l'étoile avec l'horizontale.
On tape :

```
etoile(a,r,k):={
 local  A,j,L,s1;
 A:= point(a+r*cos(2*pi/5)/cos(pi/5)*exp(i*k+i*pi/5));
```

```
 L:=NULL;
 s1:=[segment(a+r*exp(i*k),A),segment(A,a+r*exp(i*k+i*2*pi/5))];
 pour j de 0 jusque 4 faire
   L:=L,s1
   s1:=rotation(a,2*pi/5,s1);
     fpour;
 return L;
 }:;
```

La fonction `tour` dessine des pentagones sur le pourtour d'un décagone.
`tour` a comme pramètres :
`a` est l'affixe du centre du décagone,
`R` est le rayon de son cercle circonscrit,
`k` est l'angle en radian que fait un sommet du décagone avec l'horizontale.
On tape :

```
tour(a,R,k):={
  local c,r,j,L,l,ret,aet;
  c:=2*R*sin(pi/10);
  r:=c/2/sin(pi/5);
  l:=2*r*sin(2*pi/5);
  ret:=r*sin(2*pi/5)/sin(pi/5);
  aet:=R*cos(pi/10)+r*cos(pi/5)+r*cos(2*pi/5)+ret*cos(pi/5);
  L:=NULL;
  pour j de 0 jusque 9 faire
   L:=L,isopolygone(a+R*exp(i*k+i*pi*j/5)*exp(i*pi/5),
                    a+R*exp(i*k+i*pi*j/5),5);
   fpour;
  return L;
}:;
```

La fonction `touret` dessine un tour de pentagones, un tour d'étoiles, un tour fait avec tour, un tour d'étoiles et enfin un tour fait avec tour.
`touret` a les mêmes paramètres que `tour`
Pour les variables locales on a :
`R` : rayon du décagone
`r` : rayon des pentagones
`ret` : rayon des étoiles
`l` est la distance entre le sommet 0 et le sommet 2 du pentagone.
`aet` distance du centre du décagone au centre d'une étoile.
On tape :

```
touret(a,R,k):={
local c,r,j,L,l,ret,aet;
L:=NULL;
c:=2*R*sin(pi/10);
r:=c/2/sin(pi/5.);
l:=2*r*sin(2*pi/5.);
ret:=r*sin(2*pi/5.)/sin(pi/5.);
aet:=R*cos(pi/10.)+r*cos(pi/5.)+r*cos(2*pi/5.)+ret*cos(pi/5.);
```

```
pour j de 0 jusque 9 faire
 L:=L,isopolygone(a+R*exp(i*k+i*pi*j/5)*exp(i*pi/5),
                 a+R*exp(i*k+i*pi*j/5),5);
 L:=L,etoile(a+aet*exp(i*pi/10.+j*i*pi/5.),ret,pi/10.+j*pi/5.);
 L:=L,tour(a+(2*(R+c)+l)*exp(i*j*pi/5.),R,k);
 L:=L,tour(a+(4*R+3*c+l)*exp(i*j*pi/5.),R,k);
 L:=L,etoile(a+(4*R+3*c+l)*exp(i*j*pi/5.)+
          aet*exp(i*7*pi/10+j*i*pi/5.),ret,7*pi/10.+j*pi/5.);
fpour;
return L;
}:;
```

On tape :
`etoile(-35,2,0),tour(-25,2,0),touret(0,2,0)`
On obtient :



On tape pour faire un quasi-cristal :

```
R:=2;
c:=2*R*sin(pi/10);
r:=c/2/sin(pi/5.);
l:=2*r*sin(2*pi/5.);
ret:=r*sin(2*pi/5.)/sin(pi/5.);
aet:=R*cos(pi/10.)+r*cos(pi/5.)+r*cos(2*pi/5.)+ret*cos(pi/5.);
touret(0,2,0);tour(2*(R+c)+l,2,0);tour((2*(R+c)+l)*exp(i*pi/5.),2,0);
tour(4*R+3*c+l,2,0);tour((4*R+3*c+l)*exp(i*pi/5.),2,0);
tour(4*R+3*c+l+(sqrt(5)+3)*exp(2*i*pi/5.),2,0);
touret((6*R+5*c+2*l),2,0);
touret((6*R+5*c+2*l)+(6*R+5*c+2*l)*exp(2*i*pi/5.),2,0);
touret((6*R+5*c+2*l)*exp(3*i*pi/5.),2,0);
touret(2*(6*R+5*c+2*l)*cos(pi/5.)*exp(2*i*pi/5.),2,0);
tour((6*R+5*c+2*l)/cos(3*pi/10.)*exp(3*i*pi/10.)/2,2,pi/10.);
```

On obtient :



14.9 Puzzle : remplir un décagone avec des losanges

```
deca(O1,O2):={
local R,l,L,A,k;
R:=longueur(O1,O2);
l:=2*R*sin(pi/10);
L:=NULL;
A:=O2;
L:=L,A;
pour k de 1 jusque 10 faire
A:=A+2*(A-O1)*exp(3*i*pi/5)*sin(pi/10);
L:=L,A;
fpour;
return L;
}:;
penta1(O1,O2):={
local R,l,L,A,k;
R:=longueur(O1,O2);
l:=2*R*sin(pi/10);
L:=NULL;
A:=O1+2*(O2-O1)*sin(pi/10);
L:=L,A;
pour k de 1 jusque 13 faire
A:=O1+(A-O1)*exp(i*pi/5);
L:=L,A;
fpour;
return L;
}:;
penta2(O1,O2):={
local R,l,L,A,k;
R:=longueur(O1,O2);
l:=2*R*sin(pi/10);
```

```
L:=NULL;
A:=O1+4*(O2-O1)*sin(pi/10)^2;
L:=L,A;
pour k de 1 jusque 10 faire
A:=O1+(A-O1)*exp(i*pi/5);
L:=L,A;
fpour;
return L;
}:;
```

On tape :

```
point(0);
deca(point(0),point(1));
penta1(point(0),point(1));
penta2(point(0),point(1));
```

On obtient :



Cela permet de remplir un décagone avec des losanges en joignant à la souris les points du dessin ci-dessus.



Puis de colorer le dessin avec la souris :
choisir le mode quadrilatere (Polygones->quadrilatere) et comme attribut de l'objet le dessin plein (cliquer sur l'ellipse rempli et la couleur), on obtient :



Voisi les programmes de différents remplissages :

```
vasa1(O1,O2):={
local L0,L1,L2,k,P,Q;
L0:=deca(O1,O2);
L1:=penta1(O1,O2);
```

```
L2:=penta2(O1,O2);
P:=NULL;
Q:=NULL;
pour k de 0 jusque 4 faire
P:=P,polygone(L0[2k],L0[2k+1],L0[2k+2],L1[2*k+1],affichage=rempli+32*k+58);
Q:=Q,polygone(O1,L1[2k+1],L0[2k+2],L1[2k+3],affichage=rempli+60+32*k);
fpour;
return P,Q;
}
:;
vasa2(O1,O2):={
local L0,L1,L2,k,P,Q;
L0:=deca(O1,O2);
L1:=penta1(O1,O2);
L2:=penta2(O1,O2);
P:=NULL;
Q:=NULL;
pour k de 0 jusque 3 faire
P:=P,polygone(L0[2k],L0[2k+1],L0[2k+2],L1[2*k+1],affichage=rempli+k);
fpour;
pour k de 0 jusque 2 faire
Q:=Q,polygone(O1,L1[2k+1],L0[2k+2],L1[2k+3],affichage=rempli+7*k+60);
fpour;
P:=P,polygone(O1,L1[7],L2[9],L1[11],affichage=rempli+4);
Q:=Q,polygone(L0[8],L0[9],L2[9],L1[7],affichage=rempli+81);
Q:=Q,polygone(L0[9],L0[10],L1[11],L2[9],affichage=rempli+88);
return P,Q;
}
:;
vasa3(O1,O2):={
local L0,L1,L2,k,P,Q;
L0:=deca(O1,O2);
L1:=penta1(O1,O2);
L2:=penta2(O1,O2);
P:=NULL;
Q:=NULL;
P:=P,polygone(L0[0],L0[1],L0[2],L1[1],affichage=rempli+0);
pour k de 2 jusque 3 faire
P:=P,polygone(L0[2k],L0[2k+1],L0[2k+2],L1[2*k+1],affichage=rempli+k);
fpour;
Q:=Q,polygone(O1,L1[5],L0[6],L1[7],affichage=rempli+74);
P:=P,polygone(O1,L1[1],L2[3],L1[5],affichage=rempli+1);
P:=P,polygone(O1,L1[7],L2[9],L1[11],affichage=rempli+4);
Q:=Q,polygone(L0[2],L0[3],L2[3],L1[1],affichage=rempli+60);
Q:=Q,polygone(L0[3],L0[4],L1[5],L2[3],affichage=rempli+67);
Q:=Q,polygone(L0[8],L0[9],L2[9],L1[7],affichage=rempli+81);
Q:=Q,polygone(L0[9],L0[10],L1[11],L2[9],affichage=rempli+88);
return P,Q;
```

```
}
:;
vasa4(O1,O2):={
local L0,L1,L2,k,P,Q;
L0:=deca(O1,O2);
L1:=penta1(O1,O2);
L2:=penta2(O1,O2);
P:=NULL;
Q:=NULL;
P:=P,polygone(L0[0],L0[1],L0[2],L1[1],affichage=rempli+0);
P:=P,polygone(L0[2],L2[2],L2[9],L1[1],affichage=rempli+1);
P:=P,polygone(L0[3],L0[4],L0[5],L1[4],affichage=rempli+2);
Q:=Q,polygone(L0[2],L0[3],L1[4],L2[2],affichage=rempli+74);
P:=P,polygone(L0[6],L0[7],L0[8],L1[7],affichage=rempli+3);
P:=P,polygone(L0[6],L1[7],L2[9],L2[6],affichage=rempli+4);
Q:=Q,polygone(L0[5],L0[6],L2[6],L1[4],affichage=rempli+60);
Q:=Q,polygone(L1[4],L2[6],L2[9],L2[2],affichage=rempli+67);
Q:=Q,polygone(L0[8],L0[9],L2[9],L1[7],affichage=rempli+81);
Q:=Q,polygone(L0[9],L0[0],L1[1],L2[9],affichage=rempli+94);
return P,Q;
}
:;
vasa5(O1,O2):={
local L0,L1,L2,k,P,Q;
L0:=deca(O1,O2);
L1:=penta1(O1,O2);
L2:=penta2(O1,O2);
P:=NULL;
Q:=NULL;
P:=P,polygone(L0[0],L0[1],L0[2],L1[1],affichage=rempli+0);
P:=P,polygone(L0[3],L0[4],L0[5],L1[4],affichage=rempli+1);
P:=P,polygone(L0[3],L1[4],L2[6],L2[3],affichage=rempli+2);
Q:=Q,polygone(L0[2],L0[3],L2[3],L1[1],affichage=rempli+74);
P:=P,polygone(L0[6],L0[7],L0[8],L1[7],affichage=rempli+3);
P:=P,polygone(L0[6],L1[7],L2[9],L2[6],affichage=rempli+4);
Q:=Q,polygone(L0[5],L0[6],L2[6],L1[4],affichage=rempli+60);
Q:=Q,polygone(L1[1],L2[3],L2[6],L2[9],affichage=rempli+67);
Q:=Q,polygone(L0[8],L0[9],L2[9],L1[7],affichage=rempli+81);
Q:=Q,polygone(L0[9],L0[0],L1[1],L2[9],affichage=rempli+88);
return P,Q;
}
:;
```

On tape :

```
vasa1(0,1),vasa2((2*sin(pi/10)+1)*exp(i*pi/5),2*sin(pi/10)*exp(i*pi/5
vasa3((2*sin(pi/10)+1)*exp(-i*pi/5),2*sin(pi/10)*exp(-i*pi/5)),
vasa7((2*sin(pi/10)+1)*exp(2*i*pi/5)-1,2*sin(pi/10)*exp(2*i*pi/5)-1)
```

On obtient :



14.10 Les deux hélices

Ce problème aète donné aux olympiades académiques de 2005.

14.10.1 Le problème

Un avion modèle réduit possède deux hélices de même longueur qui tournent dans un même plan perpendiculaire à leurs axes, et à la même vitesse. Comment choisir la distance entre leurs axes a et l'angle de départ b des 2 hélices pour que les deux hélices puissent tourner sans se heurter ?

14.10.2 La modélisation avec `Xcas`

On suppose les hélices de centres `O1,O2` et de longueur 2. On choisit comme paramètres, la distance `a` des centres des 2 hélices, et la mesure `b` de l'angle des 2 hélices. Plus précisemment, on note l'hélice1 `A1,A2` et l'hélice2 `B1,B2` pour que l'angle $\beta = (\overrightarrow{A1, A2}, \overrightarrow{B1, B2})$ soit de mesure $b \in [0; \pi[$.
On pourra tester différentes valeurs de `a` et `b` grâce aux commandes :

```
a:=element(0..2);
b:=element(0..pi);
```

qui font apparaitre des curseurs permettant de modifier `a` ou `b`.
On va utiliser la commande `animate` qui permet de faire une animation. Il faut pour cela créer, pour chaque hélice, une séquence (de 40 ou 48 éléments) contenant les différentes positions qui seront dessinées.
On définit pour la première hélice :

```
h1:=seq(segment(exp(i*(t+pi)),exp(i*t)),t,0,2*pi,pi/20)
```

et pour la deuxième hélice :

```
h2:=
seq(segment(a+exp(i*(t+pi)),a+exp(i*t)),t,b,2*pi+b,pi/20)
```

Donc on tape pour avoir 40 positions différentes et avoir au départ `a=sqrt(2)` et `b=pi/4` :

```
h1:=seq(segment(exp(i*(t+pi)),exp(i*t)),t=0..2*pi,pi/20):;
```

```
animation(h1);
a:=element(0..2,sqrt(2));
b:=element(0..pi,pi/4);
h2:=seq(segment(a+exp(i*(t+pi)),a+exp(i*t)),
        t=b..2*pi+b,pi/20):;
animation(h2);
```

ou pour avoir 48 positions différentes et avoir au départ `a=sqrt(3)` et `b=pi/3` :

```
h1:=seq(segment(exp(i*(t+pi)),exp(i*t)),t=0..2*pi,pi/24):;
animation(h1);
a:=element(0..2,sqrt(3));
b:=element(0..pi,pi/3);
h2:=seq(segment(a+exp(i*(t+pi)),a+exp(i*t)),
        t=b..2*pi+b,pi/24):;
animation(h2);
```

14.10.3 Le raisonnement

Il y a des cas simples :

— lorsque $a \leq 1$ on est sûr que les 2 hélices se touchent quelquesoit b,

— lorsque $a > 2$ on est sûr que les 2 hélices ne se touchent pas quelquesoit b.

Il reste donc à étudier le cas $1 < a \leq 2$. Supposons qu'à un moment donné les 2 hélices se touchent : par exemple le point `A1` touche l'hélice2 en `M` avec `O2M=c` : cela forme un triangle de côtés $a, c, 1$ ($0 < c \leq 1$ et d'angle b ou $\pi - b$ opposé au côté a).
On a donc la relation :
$a^2 = 1 + c^2 - 2 * c * \cos(b) = (1 - c)^2 + 2 * c * (1 - \cos(b))$
ou la relation :
$a^2 = 1 + c^2 + 2 * c * \cos(b) = (1 - c)^2 + 2 * c * (1 + \cos(b))$
Si il y a collision c'est qu'il existe $0 < c \leq 1$ vérifiant l'une de ces 2 équations du second degré en c de discriminant $\Delta = \cos(b)^2 - 1 + a^2$.
Puisque $a > 1$, on a $a > \sin(b)$ donc $\Delta > 0$: il y a donc 2 solutions de signe contraire puisque le produit des racines vaut $1 - a^2 < 0$, donc 0 se trouve à l'intérieur des racines.
Si il y a collision c'est qu'il existe une racine comprise entre 0 et 1, donc 1 se trouve à l'extérieur des racines.
On a pour $c = 0$, $1 + c^2 - 2 * c * \cos(b) - a^2$ (resp $1 + c^2 + 2 * c * \cos(b) - a^2$) vaut $1 - a^2 < 0$ puisque $a > 1$ et,
pour $c = 1$, on a $1 + c^2 - 2 * c * \cos(b) - a^2$ (resp $1 + c^2 + 2 * c * \cos(b) - a^2$) vaut $2 - 2\cos(b) - a^2$ (resp$2 + 2\cos(b) - a^2$).
L'une de ces quantités est positive si il y a une solution entre 0 et 1, donc
$a^2 \leq 2 - 2\cos(b) - a^2 = 4 * c^2 * \sin(b/2)^2$ ou
$a^2 \leq 2 + 2\cos(b) - a^2 = 4 * c^2 * \cos(b/2)^2$
Donc si il y a collision, c'est que $a \leq 2 * \sin(b/2)$ ou $a \leq 2 * \cos(b/2)$.
Réciproquement supposons :

— $a \leq 2 * \sin(b/2)$
Il existe c entre 0 et 1 et un triangle de côtés $a, 1, c$, d'angle opposé au côté a égal à b. En effet, l'équation :

$\text{eq}(x) = x^2 - 2 * x * \cos(b) + 1 - a^2 = 0$ a une solution c comprise dans $]0; 1]$ car $\text{eq}(0) = 1 - a^2 < 0$ puisque $a > 1$ et
$\text{eq}(1) = 1 - 2 * \cos(b) + 1 - a^2 = 4 * \sin(b/2)^2 - a^2 \geq 0$ d'après l'hypothèse.

— $a \leq 2 * \cos(b/2)$
Il existe c entre 0 et 1 et un triangle de côtés $a, 1, c$, d'angle opposé au côté a égal à $\pi - b$. En effet l'équation :
$\text{eq}(x) = x^2 - 2 * x * \cos(\pi - b) + 1 - a^2 = 0$ a une solution c comprise dans $]0; 1]$ puisque $\text{eq}(0) = 1 - a^2 < 0$ car $a > 1$ et
$\text{eq}(1) = 1 + 2 * \cos(b) + 1 - a^2 = 4 * \cos(b/2)^2 - a^2 \geq 0$ d'après l'hypothèse.

Si $a \leq 2 * \sin(b/2)$ ou $a \leq 2 * \cos(b/2)$, la construction d'un tel triangle est possible ce qui prouve qu'il y a collision entre les 2 hélices.
Donc si on a $a > 2 * \sin(b/2)$ et $a > 2 * \cos(b/2)$, on est sûr que la collision n'est pas possible.
Cela veut dire que si l'on choisit $a > 2 * \max(\sin(b/2), \cos(b/2))$, il n'y aura pas de collision possible.
Par exemple :
pour $b = \pi/2$ on doit choisir $a > \sqrt{2}$,
pour $b = \pi/3$ on doit choisir $a > \sqrt{3}$.

14.11 La boite des cerfs-volants

14.11.1 La boite carrée des cerfs-volants

On veut mettre dans une boite "carrée" six cerfs-volants de tailles différentes. Ces cerfs-volants sont des quadrilatères semblables qui sont formés par 2 triangles rectangles (non isocèles) égaux ayant comme côté commun leur hypoténuse.
Soit t la valeur inférieure à $\pi/4$ de l'angle aigu de ces triangles rectangles.



On veut pouvoir ranger ces six cerfs-volants dans une boite carrée :



Pour quelle(s) valeur(s) de t cela est-il possible ?

Solution



La disposition est indépendante de l'unité choisie i.e. est invariante par homothétie.
Supposons $BM = 1$ unité.
On a :
$OA = AC = AN$ donc A est le milieu de ON.
$DQ = DE = DG$ donc D est le milieu de GQ.
$GH = HJ = HM$ donc H est le milieu de GM.
Posons : $MH = HG = \dfrac{1}{\tan(t)} = x$
On a donc aussi :

$FG = xHG = x * MH = x^2$
$GD = xFG = x^3$
$NB = xNA = xOA$
Puisque $OQMN$ est un carré on a :
$2OA = 2DH = ON$ et
$NB + 1 = 2OA$
Donc :
$2 * DH = 2(DG + GH) = 2 * (x^3 + x) = 2OA$ et
$x * OA + 1 = 2OA$
Donc :
$OA = x^3 + x$ et $(2 - x)OA = 1$
d'où l'équation : $(x^3 + x)(2 - x) = 1$
En développant on obtient :
$-x^4 + 2 * x^3 - x^2 + 2 * x - 1 = 0$
En mettant x^2en facteur on obtient : $-(x^2 + 1/x^2) + 2 * (x + 1/x) - 1 = 0$
En posant $X = x + \dfrac{1}{x}$ on a $X^2 = x^2 + 2 + \dfrac{1}{x^2}$ donc obtient l'équation :
$-X^2 + 2X + 1 = 0$.
On tape :
```
solve(-X^2+2X+1=0,X)
```
On obtient :
```
[1-sqrt(2),1+sqrt(2)]
```
donc $X = 1 + \sqrt{2}$ ou $X = 1 - \sqrt{2}$ donc :
$X = x + \dfrac{1}{x} = \dfrac{1}{\tan(t)} + \tan(t) = \dfrac{1 + \tan(t)^2)}{\tan(t)} = \dfrac{1}{\sin(t)\cos(t)} = \dfrac{2}{\sin(2t)}$
On en déduit que :
$\sin(2t) = 2/(1 + \sqrt{2}) = 2\sqrt{2} - 2$ ou
$\sin(2t) = 2/(1 - \sqrt{2}) = -2\sqrt{2} - 2$ (impossible)
Donc :
$t = \dfrac{\text{asin}(2\sqrt{2} - 2)}{2} \simeq 0.488146802076$ ou
$t = \dfrac{\pi - \text{asin}(2\sqrt{2} - 2)}{2} \simeq 1.08264952472$
Puisque $t < \pi/4 \simeq 0.785398163397$ on a $t \simeq 0.488146802076$.
Donc la 1ière solution est la figure du début de l'exercice.
Si $t > \pi/4$ la solution est obtenue à partir de la 1ière solution par une symétrie d'axe parallèle à OQ passant par A :



Le carré a alors un côté égal à $2*(x^3+x)=2x^2X$ si $x=1/\tan(t)>1$ (cas $t<\pi/4$) et égal à $2/x^3+2/x)=2X/x^2$ si $x<1$ (cas $t>\pi/4$).
On tape :
`solve(x+1/x=1+sqrt(2),x)`
On obtient :
```
[(1+sqrt(2)-(sqrt(-1+sqrt(2)*2)))/2,
(1+sqrt(2)+sqrt(-1+sqrt(2)*2))/2]
```
de valeurs approchées : `[0.53101005646,1.88320350591]` On tape :
`1.88320350591^2*2*(1+sqrt(2))` ou
`2*(1+sqrt(2))/0.53101005646^2`
On obtient :
`17.123801666`
On tape :
`longueur(O,Q)`
On obtient :
`17.123801666`
Le côté du carré a pour longueur `17.123801666*k` lorsque $BM=k$ **Remarque** Avec `Xcas` on vérifie avec, par exemple, un carré de côté 4.
On tape :

```
carre(0,8);
A:=point(4*i);
N:=point(8*i);
M:=point(8+8*i,affichage=quadrant4));
supposons(t=[0.5,0,1,0.01]);
P:=point(4*tan(t));
d:=droite(y=tan(2*t)*(x-4*tan(t)));
B:=inter_unique(droite(y=8+8*i),d);
D:=point(8+(8-4*tan(t))*tan(t)*i);
G:=point(8+(8-4*tan(t))*tan(t)*2*i);
H:=point(8+(8-4*tan(t))*(2*tan(t)+tan(t)^3)*i);
J:=projection(d,H);
```

On obtient :



On tape :

```
solve(longueur2(H,D)-16,t=0..pi/4)
```

On obtient :

```
[0.488146802076]
```

14.11.2 L'angle t des cerfs-volants vaut $\pi/6$

Le fabricant de cerfs-volants veut utiliser comme triangles rectangles des moitiés de triangles équilatéraux i.e. $t = \pi/6$.

Il veut connaitre les dimensions de la boite servant à ranger les 6 cerfs-volants ci dessous :



On suppose $BM = k$.

On a donc :

$OA = AC = AN = NB/\sqrt{3}$,

$PQ = \sqrt{3}DQ$,
$GD = ED = DQ = \sqrt{3}EF$,
$EF = GF = \sqrt{3}HG$
$HG = JH = HM = \sqrt{3}$
$BM = \sqrt{3}k$
donc $FD = DQ = 3\sqrt{3}HM = 3\sqrt{3}k$
Donc $ON = 2OA = MQ = 2(HM + DQ) = 2(\sqrt{3} + 3\sqrt{3})k = 8k\sqrt{3} \simeq 13.8564064606k$
et $OQ = NM = NB + k = OA\sqrt{3} + k = 4\sqrt{3}\sqrt{3}k + k = 13k$
La boite est donc rectangulaire et ses côtés sont $13{*}k$ et $8{*}\sqrt{3}{*}k \simeq 13.8564064606k$ lorsque $BM = k$.

Chapitre 15

Pour s'amuser en géométrie dans l'espace

15.1 Un patron du cube avec une animation

On peut montrer qu'il y a 35 hexominos non superposoables (figure formèe par 6 carrés ayant au moins une arrête en commun) et que parmi ces 35 hexominos il y en a 11 qui peuvent être le patron d'un cube.
Ce sont :



Pour faire le dessin ci-dessus on a tapé :

```
patron(a):={
```

```
  local L,L1,L2,L3,L4,L5,L6;
  L(a):=carre(a+k,a+1+k)$(k=0..3);
  L1:=L(a),carre(a+2+i,a+3+i),carre(a+2-i,a+3-i);
  a:=a+5;
  L2:=L(a),carre(a+3+i,a+4+i),carre(a+3-i,a+4-i);
  a:=a+5;
  L3:=L(a),carre(a+3+i,a+4+i),carre(a+2-i,a+3-i);
  a:=a-4i-10;
  L4:=L(a),carre(a+3+i,a+4+i),carre(a+1-i,a+2-i);
  a:=a+5;
  L5:=L(a),carre(a+3+i,a+4+i),carre(a-i,a+1-i);
  a:=a+5;
  L6:=L(a),carre(a+2+i,a+3+i),carre(a+1-i,a+2-i);
retourne L1,L2,L3,L4,L5,L6;
}:;
patron2(a):={
  local L,L1,L2,L3,L4,L5;
  L(a):=carre(a+k,a+1+k)$(k=0..2);
  L1:=L(a-1),L(a+1+i);
  a:=a+6;
  L2:=L(a),carre(a+i,a+1+i),carre(a-i,a+1-i),carre(a-1-i,a-i);
  a:=a+5;
  L3:=L(a),carre(a+1+i,a+2+i),carre(a-i,a+1-i),carre(a-1-i,a-i);
  a:=a-4i-11;
  L4:=L(a),carre(a+2+i,a+3+i),carre(a-i,a+1-i),carre(a-1-i,a-i);
  a:=a+6;
  L(a):=carre(a+k,a+1+k)$(k=0..1)
  L5:=L(a),L(a+1+i),L(a-1-i);
  retourne L1,L2,L3,L4,L5;
}:;
patron(-6+3i);
patron2(-6-5i);
```

On veut faire une animation permettant de faire le patron numéro 1 du cube.
Cette animation permet de voir comment on passe du cube au patron et du patron au cube.
On tape :

```
cubeani(t):={
local A,B,C,D,A1,B1,C1,D1,C2,D2,A3,B3,C3,D3,L;
A:=point([0,0,0]);
B:=point([1,0,0]);
C:=point([1,1,0]);C1:=point([1,1,1]);
D:=point([0,1,0]);
L:=cube(A,B,C);
A1:=point(0,-(sin(t)),cos(t));
B1:=point([1,-sin(t),cos(t)]);
C1:=point([1,cos(2t)-sin(t),sin(2t)+cos(t)]);
```

```
D1:=point([0,cos(2t)-sin(t),sin(2t)+cos(t)]);
C2:=point([1,1+sin(t),cos(t)]);
D2:=point([0,1+sin(t),cos(t)]);
C3:=point([1+sin(t),1,cos(t)]);
D3:=point([-sin(t),1,cos(t)]);
B3:=point([1+sin(t),0,cos(t)]);
A3:=point([-sin(t),0,cos(t)]);
L:=L,polygone(A1,B1,C1,D1,affichage=128+rempli):;
L:=L,polygone(A1,B1,B,A,affichage=129+rempli);
L:=L,polygone(D,C,C2,D2,affichage=130+rempli);
L:=L,polygone(C3,B3,B,C,affichage=131+rempli);
L:=L,polygone(D3,A3,A,D,affichage=132+rempli);
L:=L,segment(A1,B1);
L:=L,polygone(A,B,C,D,affichage=133+rempli);
return [L];
}:;
```

On a comme paramètres :
`A,B,C,D` sont les sommets fixes de la base du cube,
`A1,B1 C1,D1,C2,D2,A3,B3,C3,D3` sont les sommets de la face parallèle à `A,B,C,D` du cube qui ont subi des rotations : rotation d'angle t ($t = 0..\pi/2$) et
d'axe `line(A,B)` pour `A1,B1`, donc
`[B1,A1]:=rotation(line(A,B),t,[point(1,0,1),point(0,0,1)]),`
d'axe `line(A,B)` puis d'axe `line(A1,B1)` pour `C1,D1`, donc
`[C1,D1]:=rotation(line(A1,B1),t,rotation(line(A,B),a, [point(1,1,1),point(0,1`
d'axe `line(C,D)` pour `C2,D2`, donc
`[C2,D2]:=rotation(line(C,D),t,[point(1,1,1),point(0,1,1)])`
d'axe `line(B,C)` pour `B3,C3`, donc
`[B3,C3]:=rotation(line(B,C),t,[point(1,0,1),point(1,1,1)])`
d'axe `line(D,A)` pour `A3,D3` donc :
`[D3,A3]:=rotation(line(D,A),t,[point(0,1,1),point(0,0,1)])`



Puis, on tape : `L1:=seq([cubeani(t)],t=0..1.57,0.1):;`
`L2:=seq([cubeani(t)],t=1.57..0,-0.1):;`
`animation(L1,L2)`

15.2 Les pliages pour construire des deltaèdres

Selon les deltaèdres à construire, on aura besoin de pièces identiques à la pièce élémentaire et aussi de pièces symétriques à la pièce élémentaire.

15.2.1 Le rectangle de base

La pièce élémentaire se fait avec un rectangle de dimension $a \times a\sqrt{3}$.
On peut obtenir 2 rectanges de cette dimension par pliage :

— on plie une feuille A_4 de largeur AB selon la médiatrice de AB
— on fait un pli partant de B qui améne le point A en un point C situé sur cette médiatrice pour obtenir un triangle équilatéral ABC.
— On enlève la bande horizontale au dessus de C.
— On coupe selon la médiatrice pour obtenir 2 rectangles identiques ayant les bonnes proportions.

15.2.2 Le pliage du rectangle

Fabrication de la pièce élémentaire

Prendre un rectangle $ABCD$ de dimension $a \times a\sqrt{3}$.
Voilà comment on procède :

On améne A en C pour former le pli $G1G2$

On améne C en $G1$ selon le pli $C1G2$ et A en $G2$ selon le pli $G1C2$

On obtient :

On plie selon $C1G1$ pour cacher le petit triangle vert sous le rabat $CC1G2$.
On plie selon $C2G2$ pour cacher le petit triangle vert sous le rabat $AC2G1$.

On retourne la pièce et on effectue les plis : $M1G1$, $M1M2$, $M2G2$ et on obtient la pièce élémentaire formées de 4 triangles équilatèraux.

Les triangles situés à chaque bout seront appelés : **languettes** et les triangles du centre seront appelés : **pochettes**. Pour construire, on met les languettes dans les pochettes.

Fabrication de la pièce symétrique

Prendre un rectangle $ABCD$ de dimension $a \times a\sqrt{3}$.
On améne B en D pour former le pli $g1g2$ symétrique de $G1G2$ par rapport à la médiatrice de AB, puis on continue comme précédemment...

Le tétraèdre régulier

Réaliser 2 pièces : l'une identique à la pièce élémentaire (en noir) l'autre la symétrique de la pièce élémentaire (en rouge).
Pour construire le tétraèdre régulier il suffit de glisser les languettes sous la fente centrale (i.e. dans les pochettes).
Début de la construction :

15.2.3 Les Deltaèdres

Pour construire :

- Le tétraèdre régulier (4 faces) il faut 1 pièce élémentaire + 1 pièce symétrique.
- L'héxaèdre (6 faces) il faut 3 pièces identiques ou 2 pièces identiques + 1 pièce symétrique.
- L'octaèdre (8 faces) il faut 4 pièces identiques ou 2 pièces identiques + 2 pièces symétriques.
- Le décaèdre (10 faces) il faut 5 pièces identiques ou 3 pièces identiques + 2 pièces symétriques.
- L'icosaèdre (20 faces) il faut 5 pièces identiques + 5 pièces symétriques
- Le rhomboèdre dont les faces sont des losanges formés par 2 triangles équilatéraux, il faut 3 pièces identiques + 3 pièces symétriques.
- L'étoile de Kepler (60 faces constituées par les faces des 20 tétraèdres réguliers construits sur les faces de l'icosaèdre) il faut 30 pièces identiques.

15.2.4 Activité en classe

Refaire les opérations de pliage et montrer qu'une pièce élémentaire est constituée de 4 triangles équilatéraux (le pli $G1G2$ est la médiatrice de AC et ABC étant la moitié d'un triangle équilatéral, $G2$ est le centre de gravité de ce triangle équilatéral etc...).

15.3 Le rhomboèdre

15.3.1 La définition et la construction avec `Xcas`

On veut construire le rhomboèdre AB, AD, AF On construit un rhomboèdre dont les faces sont des losanges de côtés de longueur l et ayant un angle égal à a avec $0 \leq a \leq 2\pi/3$.
Pour faciliter les calculs, on choisit tout d'abord le point A sur l'axe des z et on suppose que 3 losanges $ABCD$,$ADEF$,$AFGB$ sont issus du sommet A, l'angle A de ces 3 losanges valant a et que B, D, E sont dans le plan Oxy avec B sur l'axe des x.
On tape :

```
rhomboedre0(l,a):={
local d,r,h,A,B,C,D,E,F,G;
```

```
d:=2*l*sin(a/2);
r:=d/sqrt(3);
h:=sqrt(l*l-r*r);
A:=point(0,0,h);
B:=point(r,0,0);
D:=point(-r/2,r*sqrt(3)/2,0);
F:=point(-r/2,-r*sqrt(3)/2,0);
C:=point(r/2,r*sqrt(3)/2,-h);
E:=point(r/2,-r*sqrt(3)/2,-h);
G:=point(-r/2,0,-h);
retourne parallelepipede(A,B,D,F);
}:;
```

On peut prendre un autre point de vue. pour faire le rhomboèdre AB, AC, AD, on choisit A à l'origine, B sur l'axe des x ($AB = l$), C dans le plan ABM et l'angle $\overrightarrow{AB}, \overrightarrow{AC} = a$.
On tape :

```
rhomboedre1(A,l,M,a):={
  local B,C,D,E;
  B:=point(l,0,0);
  losange(A,B,[M,a],C,D);
  E:=point(l*cos(a)*tan(a/2),l*cos(a),l*tan(a/2));
  retourne parallelepipede(A,B,C,E);
}:;
```

On fait maintenant un changement de repère, pour que A et B soient quelconques.
`changrep(A,B,M)` renvoie la matrice de passage du repère (O, i, j, k) au repère A, I, J, K avec :
$\overrightarrow{AI} = \overrightarrow{AB}/AB$, $\overrightarrow{AK} = (\overrightarrow{AI} \wedge \overrightarrow{AM})/|\overrightarrow{AI}$
$wedge\overrightarrow{AM}|$,
$\overrightarrow{AJ} = (\overrightarrow{AK} \wedge$
$overrightarrowAI)$ On tape :

```
changrep(A,B,M):={
local V,nV,W,nW,I,J,K;
V:=B-A;
nV:=sqrt(V*V);
I:=A+V/nV;
W:=cross(B-A,M-A);
nW:=sqrt(W*W);
K:=A+W/nW;
J:=A+cross(W/nW,V/nV);
retourne  normal([coordonnees(I),coordonnees(J),coordonnees(K)]);
}:;
```

On tape :

```
rhomboedre(A,B,M,a):={
```

```
  local l,C,D,E,P;
  P:=changrep(A,B,M);
  l:=longueur(A,B);
  M1:=A+tran(P)*coordonnees(M);
  retourne rhomboedre1(A,l,M1,a);
}:;
```

15.3.2 Activité en classe

Construire avec du carton un rhomboèdre : chaque groupe de 6 élèves choisit de faire un rhomboèdre à partir d'un losange donné par son angle aigu (le même losange peut donner lieu à la construction de 2 rhomboèdres).
Faire faire ensuite le dessin dans `Xcas` en prenant comme paramètre un angle du losange (on pourra suggérer de choisir la représentation de `rhomboedre0`).

15.4 Le dodécaèdre rhombique

15.4.1 La définition à partir d'un cube

Pour tracer facilement le dodécaèdre rhombique en perspective cavalière, il faut tout d'abord tracer un cube de centre O, puis tracer les 6 symétriques $S0_j$ de O par rapport au centre O_j des faces du cube ($j = 1..6$).
On joint SO_j aux quatre sommets de la face du cube de centre O_j ($j = 1..6$).
Le dodécaèdre rhombique est le solide qui a 12 faces en forme de losanges et 14 sommets qui sont les 8 sommets du cube et les 6 symétriques $S0_j$ de O par rapport au centre O_j des faces du cube ($j = 1..6$).

15.4.2 La définition et la construction avec `Xcas`

Le dodécaèdre rhombique est un polyèdre ayant 12 faces et 14 sommets. Chaque face est ègale à un même losange.
Le dodécaèdre rhombique a 6 sommets d'ordre 4 (sommet commun á 4 losanges) et 8 sommets d'ordre 3 (sommet commun á 3 losanges).
Pour se faire une idèe de sa forme, avec 8 losanges égaux de forme quelconque, on construit deux "pointes" en accolant 4 losanges. Si S et SS sont les sommets (d'ordre 4) de ces deux pointes, on ammène ensuite en coincidence les sommets opposés à S et SS. On obtient un polyèdre ayant 8 faces pleines et 4 faces vides qui peuvent être comblés par 4 losanges qui sont à priori différents des losanges initiaux. Mais pour le dodécaèdre rhombique ces losanges doivent être égaux aux losanges initiaux. Il faut donc trouver les dimensions d'une face. **Détermination des dimensions d'une face :**
On suppose qu'une face est un losange de côté l et d'angle aigu a.
Montrons que $\tan(a) = 2\sqrt{2}$.
Considérons la pointe de sommet S qui accolle 4 losanges.



Elle est constituée par les 4 losanges $SAA-1B$, SBB_1C, SCC_1D, SDD_1A où $SABCD$ est une pyramide de hauteur h et de base carrée $ABCD$ ($(AC/2)^2 = l^2-h^2$ et $AB = AC/\sqrt{2}$) et où A_1 est le symétrique de S par rapport au milieu de AB, B_1 est le symétrique de S par rapport au milieu de BC etc...
On a : $\widehat{A_1BB_1} = \widehat{ASC}$ On veut avoir : $\widehat{A_1BB_1} = \pi - a$ Dans le triangle SAB avec le théorème d'Al-Kashi on a :
$AB^2 = SA^2 + SB^2 - 2SA * SB\cos(\widehat{ASB}) = 2l^2 - 2l^2\cos(a)$ Dans le triangle SAC avec le théorème d'Al-Kashi on a :
$AC^2 = SA^2 + SC^2 - 2SA * SC\cos(\widehat{ASC}) = 2l^2 + 2l^2\cos(a)$ Donc :
$AC^2+AB^2 = 2l^2+2l^2\cos(a)+2l^2-2l^2\cos(a) = 4l^2$ Or $AC^2 = AB^2+BC^2 = 2AB^2 = 4(l^2-h^2)$ Donc :
$AC^2 + AB^2 = 4l^2 = 4l^2 - 4h^2 + 2l^2 - 2h^2$
Soit :
$l^2 = 3h^2$, $AC^2 = 4l^2 - 4h^2 = 4l^2(1-1/3) = 8l^2/3$, $AB^2 = AC^2/2 = 4l^2/3$
D'où $\sin(a/2) = AB/(2l) = 1/\sqrt{3}$ et $\cos(a/2) = \sqrt{2}/\sqrt{3}$.
Puisque $4l^2/3 = AB^2 = 2l^2 - 2l^2\cos(a)$, on a :
$\cos(a) = 1 - AB^2/(2l^2) = 1 - 2/3 = 1/3$ $\sin(a) = \sqrt{2}\frac{2}{3}$ et donc

$$\tan(a) = 2\sqrt{2}$$

Pour avoir une valeur de a en radians, on tape :
`evalf(atan(2*sqrt(2))`
On obtient : `1.23095941734` radians
Pour avoir une valeur de a en degrés, on tape :
`evalf(atan(2*sqrt(2))*180/pi)`
On obtient : `70.5287793655` degrés.
On tape pour faire le dessin d'une "pointe" :

```
dodecarhomb():={
```

```
S:=point(0,0,1/sqrt(3));
A:=point(sqrt(2/3),0,0);
B:=point(0,sqrt(2/3),0);
C:=point(-sqrt(2/3),0,0);
D:=point(0,-sqrt(2/3),0);
polyedre(S,A,B,C,D);
A1:=symetrie(milieu(A,B),S);
B1:=symetrie(milieu(B,C),S);
C1:=symetrie(milieu(C,D),S);
D1:=symetrie(milieu(D,A),S);
polygone(A,A1,B,B1,C,C1,D,D1,A);
}:;
```

On tape pour faire le dessin du dodécaèdre rhombique :

```
dodecarhomb():={
S:=point(0,0,1/sqrt(3));
A:=point(sqrt(2/3),0,0);
B:=point(0,sqrt(2/3),0);
C:=point(-sqrt(2/3),0,0);
D:=point(0,-sqrt(2/3),0);
polyedre(S,A,B,C,D);
A1:=symetrie(milieu(A,B),S);
B1:=symetrie(milieu(B,C),S);
C1:=symetrie(milieu(C,D),S);
D1:=symetrie(milieu(D,A),S);
polygone(A,A1,B,B1,C,C1,D,D1,A);
SS:=point(0,0,-3/sqrt(3));
AA:=point(sqrt(2/3),0,-2/sqrt(3));
BB:=point(0,sqrt(2/3),-2/sqrt(3));
CC:=point(-sqrt(2/3),0,-2/sqrt(3));
DD:=point(0,-sqrt(2/3),-2/sqrt(3));
polyedre(SS,AA,BB,CC,DD);
polygone(AA,A1,BB,B1,CC,C1,DD,D1,AA);
}:;
```

On obtient en cachant les faces :



15.4.3 Activité en classe

Trouver les dimensions du losange qui sera les faces du dodécaèdre rhombique. Faire la construction d'un tel losange et construire le polyèdre avec du carton puisavec `Xcas`.

15.4.4 Le graphe du dodécaèdre rhombique

On tape :

```
L:=[0,5,5+5*i,5*i,1+i,3+i,3+3*i,1+3*i,2+2*i,4+2*i,
    4+4*i,2+4*i,2+3*i,3+2*i]];
affichage(point(L[k])$(k=0..13),1+point_point+ epaisseur_point_3);
carre(L[0],L[1]);
carre(L[4],L[5]);
carre(L[8],L[9]);
segment(L[0],L[8]);
segment(L[1],L[5]);
segment(L[1],L[9]);
segment(L[2],L[6]);
segment(L[3],L[11]);
segment(L[3],L[7]);
```

On obtient :



Le squelette ou le graphe de ce dodécaèdre rhombique n'est pas hamiltonien i.e. il n'existe pas de chemin qui en empruntant les arêtes, passe une fois et une seule par chaque sommet. En effet chaque sommet d'ordre 4 n'est relié qu'à des sommets d'ordre 3 et chaque sommet d'ordre 3 n'est relié qu'à des sommets d'ordre 4. Donc le chemin ne peut passer qu'en alternant sommet d'ordre 3, sommet d'ordre 4, sommet d'ordre 3 etc...Comme il y a 6 sommets d'ordre 4 et 8 sommets d'ordre 3, un chemin passant une fois et une seule par chaque sommet n'existe pas.
La formule de Descartes dit un polyèdre convexe qui a F faces polygomales, A arêtes, S sommets alors $F+S-A=2$ (ici on a 12 faces 14 sommets et 24 arêtes et on a bien 12+14-24=2). En effet on suppose que ce polyèdre est inscrit dans une sphère sinon on le déforme pour que ce soit le cas. En rajoutant 1 sommet on obtient $N-1$ faces supplémentaires, 1 sommet supplémentaires et N arêtes supplémentaires donc $F+S-A=cste$. Pour un tétraèdre on a 4 faces, 4 sommet et 6 arêtes donc $cste=2$

15.4.5 Le patron du dodécaèdre rhombique

```
losange_dode(A,B):={
  local C,D,l,a;
  a:=atan(2*sqrt(2));
  D:=rotation(A,a,B);
  C:=D+(B-A);
  retourne quadrilatere(A,B,C,D);
  }:;
patron_dode(A,B):={
local C,D,E,F,G,L,a,b,los;
  a:=atan(2*sqrt(2));
```

```
  b:=pi-a;
  D:=rotation(A,a,B);
  C:=D+(B-A);
los:=quadrilatere(A,B,C,D);
E:=rotation(C,a,A);
F:=rotation(C,a,B);
G:=rotation(F,pi-3*a,C);
L:=los,rotation(A,-atan(2*sqrt(2)),los),rotation(C,atan(2*sqrt(2)),los),losar
retourne G,[L];
}:;
patron_dodeca(A,B):={
local C,D,E,F,G,L,L1,L2,a,los;
  a:=atan(2*sqrt(2));
  F,L:=patron_dode(A,B);
  G,L1:= patron_dode(F,F+(B-A));
  L2:= patron_dode(G,G+(B-A))[1];
L:==op(L),op(L1),op(L2)
  retourne L;
}:;
```

On tape :

`a:=atan(2*sqrt(2));patron_dodeca(point(0),point(exp(-i*a/2)))`

On obtient :



On peut aussi faire tracer ce patron par la tortue : avec la tortue il faut travailler en degrés.

On tape :

`evalf(atan(2*sqrt(2))*180/pi);`

On obtient la valeur approchée de a en degrés :

`70.5287793655` soit $70^o31.72676193'$, soit $70^o31'43.6057158''$

On tape :

```
los(l,a):={
repete(2,avance(l),tourne_gauche(a),avance(l),
       tourne_gauche(180-a));
}:;
loss(l,a):={
repete( 3,los(l,180-a),tourne_gauche(180-a));
tourne_droite(180-3*a);avance(l);tourne_gauche(180-2*a);
```

```
los(l,a);avance(l);tourne_gauche(a);
avance(l);tourne_gauche(180-2*a);
saute(l);tourne_droite(-3*a);
}:;
patron(l):={
local a;
a:=evalf(atan(2*sqrt(2))*180/pi);
tourne_droite(180-3*a/2);
repete(3,loss(l,a));
}:;
```

On tape :
`patron(30)`
On obtient :



15.5 Le triacontaèdre rhombique

15.5.1 La définition et la construction avec `Xcas`

Le triacontaèdre rhombique est un polyèdre qui a 30 faces. Ses faces sont des losanges d'or (le rapport des diagonales est égal au nombre d'or) de petit angle atan(2) radians soit environ 63.4349488229 degrés ($\simeq\ 63^o25'$) et de grand angle $\pi -$ atan(2) radians soit environ 116.565051177degrés ($\simeq\ 116^o34'$).
Étymologie : du grec triaconta "trente" et rhombos "losange" (polyèdre à trente faces en losange).
Il a été étudié par Catalan en 1862.
Il a des 32 sommets (20 sommets de degré 3 i.e. commun à 3 losanges, 12 sommets de degré 5 i.e. commun à 5 losanges)) et 60 arêtes de longueur a. Son angle dièdre vaut $4\pi/5rad = 144^o$.

La construction d'une face

Construction d'un losange $ABCD$ tel que $\overrightarrow{AB}, \overrightarrow{AD} =$ atan(2) et $AB = \sqrt{5}$:
On porte sur AB le point M tel que $AM = 1$ et sur la perpendiculaire en M à AB, on définit le point N tel que $MN = 2$ et l'angle $\overrightarrow{AM}, \overrightarrow{AN} > 0$.

Alors l'angle $\overrightarrow{AM}, \overrightarrow{AN} = \text{atan}(2)$.
On finit ensuite la construction du losange $ABCD$.
Le segment AN a pour longueur $\sqrt{5}$ donc on peut aussi construire le losange $APQN$ en portant sur AB le point P tel que $AP = AN = \sqrt{5}$.
On tape dans un niveau de géométrie 2d :

```
A:=point(0);
B:=point(5);
M:=point(1);
N:=point(1+2i);
losange(A,B,angle(A,M,N),C,D);
P:=point(sqrt(5));
losange(A,P,atan(2),Q);
```

On obtient :



L'angle dièdre du triacontaèdre

Soit D un sommet de degré 3 du triacontaèdre. Ce sommet est relié à un triangle équilatèral ABC par les 3 arêtes DA, DB, DC.
On pose $a = \text{atan}(2)$. donc l'angle $\widehat{ADB} = \pi - \text{atan}(2) = \pi - a$.
On choisit un repère tel que :
ABC est dans le plan Oxy,
D se projette en O et
$A = (1, 0, 0)$.
Donc $AB = AC = BC = \sqrt{3}$
Si $l = AD$, on a, en considérant le triangle ABD :
$AB^2 = 2 * l^2 - 2 * l^2 \cos(\pi - a)$ donc
$2 * l^2 + 2 * l^2 \cos(a) = 3$ Donc $l^2 = 3/2/(1 + cos(a))$
On tape :

`cos(atan(2))`
On obtient : `(sqrt(5))/5`
On tape :
`normal(solve(2*x-2*x*cos(pi-atan(2))-3=0))`
On obtient : `[(-3*sqrt(5)+15)/8]`
Donc $l^2 = \frac{-3\sqrt{5}+15}{8}$
La hauteur $h = DO$ du tétraèdre $DABC$ vaut donc :
$h^2 = l^2 - 1 = \frac{7-3\sqrt{5}}{8}$ On cherche à évaluer l'angle des plans ADB et ADC.
On tape dans un niveau de géométrie 3d :

```
A:=point(1,0,0);
B:=point(-1/2,sqrt(3)/2,0);
C:=point(-1/2,-sqrt(3)/2,0);
a:=atan(2);
l2:=normal(3/2/(1+cos(a)));
h:=simplify(sqrt(l2-1));
//h:=(3-sqrt(5))/4;
D:=point(0,0,h);
P:=plan(A,B,D);
Q:=plan(A,C,D);
angle(P,Q);
```

On obtient :
`acos((-(sqrt(5))-1)/4)`
On sait que $\cos(4\pi/5) = 2 * \cos(\pi/5)^2 - 1$ et que :
$\cos(2\pi/5) = \frac{\sqrt{5}-1}{4}$ donc
$\cos(4\pi/5) = \frac{3-\sqrt{5}}{4} - 1 = \frac{-\sqrt{5}-1}{4}$
Donc l'angle dièdre du triacontaèdre est de $\dfrac{4\pi}{5}$

Le patron du triacontaèdre

On tape :

```
losange_triac(A,B):={
  local C,D,l,a;
  a:=atan(2);
  D:=rotation(A,a,B);
  C:=D+(B-A);
  retourne quadrilatere(A,B,C,D);
  }:;
patron_triac1(A,B):={
local C,D,E,F,a,b,los;
  a:=atan(2);
  b:=pi-a;
  D:=rotation(A,a,B);
  C:=D+(B-A);
los:=quadrilatere(A,B,C,D);
los:=los,losange_triac(A,D);
E:=rotation(A,-2*a,D);
```

```
los:=los,losange_triac(A,E);
F:=rotation(B,2*a,A);
retourne F,[los];
}:;
patron_triac2(A,B):={
local C,D,E,F,a,b,los;
  a:=atan(2);
  b:=pi-a;
  D:=rotation(A,a,B);
  C:=D+(B-A);
los:=quadrilatere(A,B,C,D);
los:=los,losange_triac(A,D);
E:=rotation(A,-2*a,D);
los:=los,losange_triac(A,E);
F:=rotation(D,b,C);
retourne F,[los];
}:;
patron_triaconta(A,B):={
local C,D,E,F,G,l,L1,L2,a,b,j,los;
  a:=atan(2.);b:=pi-a;
  los:=NULL;
  l:=B-A;
  pour j de 1 jusque 5 faire
  F,L1:=patron_triac1(A,B);
  G,L2:= patron_triac2(F,F+l);
  los:=los,L1,L2;
  A:=G;
  B:=A+l;
  fpour;
  retourne los;
}:;
```

On tape :

```
a:=atan(2.);patron_triaconta(point(0),point(exp(-i*a/2)))
```

On obtient :



La construction du triacontaèdre avec du carton

Il faut débuter la construction du triacontaèdre en accolant 5 losanges pour former une "pointe" ayant pour base un pentagone régulier : c'est la calotte du dessus. Puis, on rajoute 5 losanges pour combler les vides (cf dessin). On forme ainsi le début de 5 nouvelles "pointes". Faire la même chose pour faire la calotte du dessous. Il vous reste alors 10 losanges pour relier les 2 calottes et terminer les amorces des 5 "pointes" du dessus et les 5 "pointes" du dessous.
En tout, le triacontaèdre a 12 "pointes". Chaque losange appartient à 2 "pointes",le triacontaèdre est donc composé de 12*5/2=30 losanges.
On tape dans un niveau de géométrie 2d pour avoir la vue de dessus d'une calotte :

```
losangetria(A,B):=losange(A,B,atan(2));
vuedessus (A,B):={
local C,D,L,j;
L:=NULL;
pour j:=1 jusque 5 faire
L:=L,losange(A,B,2*pi/5,C,D);
L:=L,losange(D,C,4*pi/5);
B:=D;
fpour;
retourne L;
}:;
```

On tape :
`vuedessus(0,-2*i)`
On obtient :



La construction du triacontaèdre avec `Xcas`

On peut faire le dessin dans l'écran de géométrie 3D.
On prend un losange de côté $a\sqrt{5}$. Pour faire le dessin avec `Xcas`, on doit faire quelques calculs.
On rappelle que :

$$\cos(\frac{2\pi}{5}) = \frac{(\sqrt{5}-1)}{4} \qquad \cos(\frac{2\pi}{5})^2 = \frac{(3-\sqrt{5})}{8}$$
$$\cos(\frac{\pi}{5}) = \frac{(\sqrt{5}+1)}{4} \qquad \cos(\frac{\pi}{5})^2 = \frac{(3+\sqrt{5})}{8}$$
$$\sin(\frac{2\pi}{5})^2 = \frac{(5+\sqrt{5})}{8} \qquad \sin(\frac{\pi}{5})^2 = \frac{(5-\sqrt{5})}{8}$$

En effet, $1+2\cos(\frac{2\pi}{5})+2\cos(\frac{4\pi}{5})=0$ donc $\cos(\frac{2\pi}{5})$ est la solution positive de $4X^2+2X-1=0$.

— Calcul de la diagonale BD^2 lorsque $AB^2=5a^2$.
  $BD^2=(AB-a)^2+(2a)^2=a^2(\sqrt{5}-1)^2+4a^2)=a^2(10-2\sqrt{5})=2a^2(5-\sqrt{5})$
— Calcul de la diagonale AC^2 lorsque $AB^2=5a^2$.
  $AC^2=(AB+a)^2+(2a)^2=a^2(\sqrt{5}+1)^2+2^2)=a^2(10+2\sqrt{5})=2a^2(5+\sqrt{5})$
— Calculons de la hauteur $h=AO$ issue de A de la pyramide de sommet A et de base le pentagone de côté BD. Cette pyramide a pour arête $AB=a\sqrt{5}$ et calculons le rayon $r=OB$ du cercle circonscrit à ce pentagone.



  On a :
  $BD=2r\sin(\pi/5)$
  On sait que $\cos(2\pi/5)=\frac{(\sqrt{5}-1)}{4}$ donc on tape :
  `normal(1-((sqrt(5)-1)/4)^2)`
  On obtient la valeur de $\sin(2\pi/5)^2$:
  `(sqrt(5)+5)/8`
  On sait que $\cos(\pi/5)=\frac{(\sqrt{5}+1)}{4}$ donc on tape :
  `normal(1-((sqrt(5)+1)/4)^2)`
  On obtient la valeur de $\sin(\pi/5)^2$:
  `(5-sqrt(5))/8`
  $BD^2=4r^2\sin(\pi/5)^2=r^2\frac{(5-\sqrt{5})}{2}=a^2(10-2\sqrt{5})$
  Donc $r^2=4a^2$
  $h^2=AB^2-r^2=5a^2-4a^2=a^2$
  Donc si $AB=a\sqrt{5}$ on a $h=AO=a$ et $OB=OC=r=2a$
— Calculons les angles $b=\widehat{MAO}$ et $c=\widehat{OAC}$.
  On a :
  $\tan(b)=\frac{2*OB\cos(\pi/5)}{OA}=2\cos(\pi/5)=\frac{(\sqrt{5}+1)}{2}$ et $\tan(c)=\frac{OC}{OA}=2$
  Donc l'angle $\widehat{MAC}=b+c$. Dans la suite on appellera cet angle : "angle au sommet" de la pyramide de sommet A et on dira que AC est l'arête opposée à AM (M etant le milieu de BD).

— Traçons le losange ABA_1D. $A1$ est aussi le sommet d'une pyramide à base pentagonale. Cette pyramide a le losange ABA_1D en commun avec la pyramide de sommet A. Son "angle au sommet" vaut donc aussi $b+c$.

Le plan médiateur de BD contient le segment AA_1 ainsi que les arêtes opposées AC et A_1C_1 où A_1C_1 l'arête opposée à A_1M pour la pyramide de sommet A_1. On va montrer que A_1C_1 est parallèle à AO (O pied de la fauteur de la pyramide de sommet A).
Calculons $\widehat{C_1A_1A}+\widehat{MAO}=b+c+b=2b+c$.
On a $\tan(2b+c)=(2\tan(b)/(1-\tan(b)^2)+\tan(c))/(1-\tan(2b)\tan(c))$.
Donc $\tan(2b+c)=(2\tan(b)+\tan(c)-\tan(c)\tan(b)^2)/D$ Calculons le numérateur on a $\tan(b)^2=\sqrt{5}+3$ donc :
$(2\tan(b)+tan(c)-\tan(b)^2\tan(c))=\sqrt{5}+1+2-\sqrt{5}-3=0$
Donc $\tan(2b+c)=0$. Donc si AO est vertical, il en est de même de A_1C_1.
Ainsi les losanges qui relient 2 calottes de sommets opposés S_1S_2 ont des côtés parrallèles à S_1S_2.



On tape pour voir les 2 calottes et la tranche du milieu :

```
S1:=point(0,0,5.5):;S1;
S2:=point(0,0,-5.5):;S2;
P1:=isopolygone(point([0,0,4.5]),point([2.,0,4.5]),point([0,1.,4.5]),-
P1:;
P2:=isopolygone(point([0,0,-4.5]),point([2*cos(pi/5.),2*sin(pi/5.),-4
point([0,1.,-4.5]),-5):;P2:;
LS1:=sommets(P1):;LS1:=op(LS1),LS1[0]:;
LS2:=sommets(P2):;LS2:=op(LS2),LS2[0]:;
L1:=losange(S1,LS1[0],LS1[1],C1):;
losange(S1,LS1[1],LS1[2],C2):;
LL1:=losange(LS1[1],C1,C2,C3):;
affichage(rotation(droite(S1,S2),2*k*pi/5,L1)$(k=0..4), 1);
affichage(rotation(droite(S1,S2),2*k*pi/5,LL1)$(k=0..4), 1);
```

```
L2:=losange(S2,LS2[0],LS2[1],D1):;
losange(S2,LS2[1],LS2[2],D2):;
LL2:=losange(LS2[1],D1,D2):;
affichage(rotation(droite(S1,S2),-2*k*pi/5,L2)$(k=0..4), 4);
affichage(rotation(droite(S1,S2),-2*k*pi/5,LL2)$(k=0..4), 4);
C4:=translation([0,0,-2.],C3):;
C5:=translation([0,0,-2.-sqrt(5.)],C3):;
C6:=translation([0,0,-2.],C1):;
losange(LS1[1],C1,C2,C3):;
C7:=translation([0,0,-2.],C2):;
LL4:=losange(C4,C5,C7):; LL3:=losange(C4,C6,C5):;
affichage(rotation(droite(S1,S2),2*k*pi/5,LL4)$(k=0..4),0);
affichage(rotation(droite(S1,S2),2*k*pi/5,LL3)$(k=0..4),0);
```

On obtient les 2 calottes et la tranche du milieu :







On peut remarquer que si A et B sont des sommets opposés et si P_A et P_B sont les plans perpendiculaires à AB passant respectivement par A et B alors les autres sommets des losanges sont disposés sur des cercles dont les 6 plans sont perpendiculaires à AB et les 8 plans (ces 6 plans + P_A et P_B) sont distants de A de $(0, a, 2a, 3a, a(\sqrt{5}+2), a(\sqrt{5}+3)), a(\sqrt{5}+4), a(\sqrt{5}+5)$. C'est à dire que $AB = a(5+\sqrt{5})$ i.e. les sommets opposés sont distants de $a(5+\sqrt{5})$.
Ses différents sommets sont sur des cercles de rayon : 0, $2a$, $4a\cos(\pi/5)$, $4a\cos(\pi/5)$, $4a\cos(\pi/5)$, $4a\cos(\pi/5)$, $2a$, 0.



Pour faire le triacontaèdre il reste à rassembler les 2 calottes et la tranche du milieu. Pour cela il suffit de changer les cotes des objets géométriques afin de rassambler les 3 images ou bien on taper :

```
//on est en dimension 2
P1:=isopolygone(point(0),2,-5):;
LS1:=op(sommets_abca(P1)):;
P2:=isopolygone(point(0),2+2*exp(i*2*pi/5),-5):;
LS2:=op(sommets_abca(P2)):;
//on passe en dimension 3
LS1:=append(evalf(coordonnees(LS1[k])),0)$(k=0..5);
LS2:=append(evalf(coordonnees(LS2[k])),-1)$(k=0..5):;
P:=seq(polygone(point(0,0,1),LS1[k],LS2[k],LS1[k+1]) ,k,0,4):;
P;
C:=NULL:;LS2:=evalf(LS2[4]),LS2:;
L:=seq(losange(LS1[k],LS2[k],LS2[k+1],C[k]),k,0,4):;L;
D:=(translation([0,0,-sqrt(5)],C[k]))$(k=0..4):;
F:=(translation([0,0,-sqrt(5)],point(LS2[k])))$(k=0..4):;
polygone(LS2[1],C[0],D[0],F[1]);
polygone(point(LS2[k+1]),C[k],D[k],point(F[k+1]))$(k=0..3):;
polygone(LS2[1],F[1],D[1],C[1]);
polygone(LS2[0],C[4],D[4],F[0]);
polygone(LS2[k],F[k],D[k],C[k])$(k=0..4);
polygone(LS2[k+1],C[k],D[k],F[k+1])$(k=0..3);
LL4:=losange(F[0],D[4],D[0],G4):;G:=NULL:;
L4:=(losange(F[k+1],D[k],D[k+1],G[k])$(k=0..3)),LL4:;
L4;G:=G4,G:;
L5:=losange(D[k],G[k],G[k+1])$(k=0..3);
LL5:=losange(D[4],G[4],G[0],AAA):;
coordonnees(AAA);L5:=L5,LL5[0]:;
```

```
affichage(L5[k],1+epaisseur_ligne_2)$(k=0..4);
```



Les coordonnées de AAA sont $(0, 0, -4 - \sqrt{5})$. Donc si une pointe a comme coordonnéees $0, 0, 1$ les coordonnées de la pointe opposée est $[0, 0, -4 - \sqrt{5}]$.
Un triacontaèdre de côté $\sqrt{5}$ a comme hauteur $5 + sqrt(5)$: la bande intermédiaire a des sommets sur une verticale de côté $\sqrt{5}$ mais ces sommets sont alternativement sur des plans horizontaux distants de 1.

Le graphe du triacontaèdre

Le triacontaèdre a 30 faces qui sont des losanges d'or, 32 sommets et 30 arêtes.

Exercice (niveau 3-ième)
Le triacontaèdre a 30 faces qui sont des losanges d'or et 32 sommets qui sont soit d'ordre 3 soit d'ordre 5 (i.e. soit communs à 3 faces, soit communs à 5 faces).
Combien y-a-t-il de sommets d'ordre 3 et de sommets d'ordre 5 ?

Soit x le nombre de sommets d'ordre 3 et y le nombre de sommets d'ordre 5.
On a à résoudre le système :
$x + y = 32$ et $3 * x + 5y = 30 * 4$ qui est équivalent à
$x = 32 - y$ et $3 * 32 + 2y = 4 * 30$ soit
$x = 32 - y$ et $y = 60 - 48$
Donc $x = 20$ et $y = 12$.
Ou bien avec `Xcas`, on tape :
`linsolve([x+y=32,3x+5y=120],[x,y])` On obtient `[20,12]`

Le graphe du triacontaèdre
On tape :

```
LA:==[-6,-4,-2,-1,1,2,4,6,-6i,-4i,-2i,-i,i,2i,4i,6i];
LB:=[3+i,1+2*i,1+3*i,2+3*i,-3+i,-1+2*i,-1+3*i,-2+3*i,
      -3-i,-1-2*i,-1-3*i,-2-3*i,3-i,1-2*i,1-3*i,2-3*i];
affichage(point(LA[k]),1+ epaisseur_point_3+point_point)$(k=0..15);
affichage(point(LB[k]),1+ epaisseur_point_3+point_point)$(k=0..15);
```

```
polygone(LA[0],LA[8],LA[7],LA[15]);
polygone(LA[1],LA[9],LA[6],LA[14]);
polygone(LA[1],LA[8],LA[6],LA[15]);
polygone(LA[4],LA[5],LB[1],LA[12]);
polygone(LA[3],LA[2],LB[5],LA[12]);
polygone(LA[3],LA[2],LB[9],LA[11]);
polygone(LA[4],LA[5],LB[13],LA[11]);
segment(LB[0],LA[5]) ,segment(LB[12],LA[5]);
segment(LB[4],LA[2]) ,segment(LB[8],LA[2]);
segment(LB[2],LB[3]), segment(LB[2],LB[1]);
segment(LB[2],LA[13]);
segment(LB[6],LB[7]), segment(LB[6],LB[5]);
segment(LB[6],LA[13]);
segment(LB[10],LB[11]), segment(LB[10],LB[9]);
segment(LB[10],LA[10]);
segment(LB[14],LB[15]), segment(LB[14],LB[13]);
segment(LB[14],LA[10]);
segment(LA[0],LA[1]), segment(LA[6],LA[7]);
segment(LA[8],LA[9]), segment(LA[14],LA[15]);
 segment(LA[10],LA[11]), segment(LA[12],LA[13]);
```

On obtient :



Le squelette de ce triacontaèdre n'est pas hamiltonien i.e. il n'existe pas de chemin qui en empruntant les arêtes, passe une fois et une seule par chaque sommet. En effet chaque sommet d'ordre 5 n'est relié qu'à des sommets d'ordre 3 et chaque sommet d'ordre 3 n'est relié qu'à des sommets d'ordre 5. Donc le chemin ne peut passer qu'en alternant sommet d'ordre 3, sommet d'ordre 5, sommet d'ordre 3 etc...Comme il y a 12 sommets d'ordre 5 et 20 sommets d'ordre 3, un chemin passant une fois et une seule par chaque sommet n'existe pas.

15.5.2 Activité en classe

Faire faire le patron du losange de base de côté $\sqrt{5}$ aux élèves pour obtenir 30 losanges (c'est parfait pour une classe de 30 élèves !).
Faire le montage avec du scotch pour obtenir le triacontaèdre.
Ou encore faire le patron de 5 losanges accolés pour pouvoir faire facilement une "pointe". il faut 12 "pointes" : le montage sera plus facile car le triacontaèdre final aura une epaisseur de 2 feuilles.
Choisir un repère orthonormé et faire calculer les coordonnées de chaque sommet du triacontaèdre : c'est un bon exercice qui utilise la géométrie analytique (en particulier, trouver les coordonnées de la somme de 2 vecteurs, l'utilité du produit scalaire...) et un peu de géométrie pure.
Voici une coupe du triacontaèdre par un plan passant par 2 sommets opposés et par une "pointe".
On tape :

```
S:=point(i);
A:=point(-2);
A1:=point(2*cos(pi/5));
S1:=symetrie(A1,S);
segment(A,S);
segment(S1,S);
d:=parallele(S1,droite(A,S)):;
O1:=inter_unique(d,droite(x=0));
segment(S1,O1);
longueur(S,O1);
S2:=S1-i*sqrt(5);
B:=point(-i-4*cos(pi/5)^2);
B1:=symetrie(B,A);
B2:=B1-i*sqrt(5);
S3:=point(-i*(sqrt(5)+4));
droite(milieu(B2,S3),2+milieu(B2,S3)));
M:=milieu(B2,S3);
N:=point(2-i*(3+sqrt(5)));
segment(S1,B2);
segment(S2,B1);
polygone(S,A,B1,B2,S3,N,S2,S1);
```

On obtient :



Un exemple d'exercice :
On a dit : "Il faut débuter la construction du triacontaèdre en accolant 5 losanges dont un angle vaut atan(2) pour former une "pointe" (la calotte du dessus) ayant pour base un pentagone régulier.
Puis, on rajoute 5 losanges pour combler les vides.
Pmontrons les vides peuvent être comblés par un losange ayant des côtés de longueurs $\sqrt{5}$ et comme angles $\alpha = \text{atan}(2)$ et $\pi - \alpha$:
On tape :

```
S:=point(0,0,1);A:=point(2,0,0),B:=point(2*cos(4*pi/5),2*sin(4*pi/5),
```

Alors le produit scalaire de $\overrightarrow{SA}.\overrightarrow{SB}$ vaut :
$4\cos(4\pi/5) + 1 = 1 - 4\cos(\pi/5) = 1 - (1 + \sqrt{5}) = -\sqrt{5} = 5 * \cos(x)$.
Donc $\cos(x) = -\sqrt{5}/5$. Puisque $\cos(\alpha) = \sqrt{5}/5$ on a $\cos(x) = -\cos(\alpha)$.
Donc l'autre angle x du losange vaut $x = \pi - \alpha$
Donc le vide peut être comblé par un losange d'angles $\alpha = \text{atan}(2)$ et $\pi - \alpha$.

15.6 La forme géométrique d'une molécule dans l'espace

15.6.1 Présentation du problème

Ce qui suit a été inspiré d'une épreuve à l'oral de l'agrégation externe de Mathématiques session 2008 dont voila le lien :
http ://agreg.dnsalias.org/Textes/pub2008-C2.pdf
Les différentes configurations possibles de chaque molécule formant un cycle de 6 atomes (comme le cyclohexane et le glucose) sont déterminées par la longueur de ses liaisons et par les angles que doivent faire entre elles les différentes liaisons.
Nous étudions ici la forme des molécules du cyclohexane et du glucose. Nous allons commencer tout d'abord par un lemme utile pour l'étude de ces 2 cas.

15.6.2 Un lemme

Pour étudier la forme de la molécule du cyclohexane, on a besoin de résoudre le système en u, v, w de la forme :

$$R(u, v) = 0, R(v, w) = 0, R(u, w) = 0$$

lorsque $R(u, v) = \alpha u^2 v^2 + \beta(u^2 + v^2) + \gamma uv + \delta$
et
pour étudier la forme de la molécule du glucose, on a besoin de résoudre le système en u, v, w de la forme :

$$P(u, v) = 0, R(v, w) = 0, R(u, w) = 0$$

lorsque $P(u, v) = \alpha_1 u^2 v^2 + \beta_1(u^2 + v^2) + \gamma_1 uv + \delta_1$ et
$R(u, w) = \alpha u^2 w^2 + \beta u^2 + \gamma uw + \eta w^2 + \delta$.
On commence par chercher à résoudre le système en u, v, w de la forme :

$$P(u, v) = 0, R(v, w) = 0, R(u, w) = 0$$

lorsque $P(u, v) = \alpha_1 u^2 v^2 + \beta_1(u^2 + v^2) + \gamma_1 uv + \delta_1$ et
$R(u, w) = \alpha u^2 w^2 + \beta u^2 + \gamma uw + \eta w^2 + \delta$.

Pour cela on cherche u et v en fonction de w pour que :
$R(v, w) = 0, R(u, w) = 0$
On a donc à résoudre tout d'abord l'équation du second degré en w :
$R(x, w) = \alpha x^2 w^2 + \beta x^2 + \gamma wx + \eta w^2 + \delta = (\alpha w^2 + \beta)x^2 + \gamma wx + \eta w^2 + \delta$.
En posant :
$a = \alpha w^2 + \beta$
$b = \gamma w$
$c = \eta w^2 + \delta$
On a donc à résoudre l'équation du second degré :
$R(x, w) = ax^2 + bx + c = 0$ où a, b, c dépendent de w.
Cette équation a 2 solutions x_1 et x_2 qui dépendent de w.
Donc on peut avoir :
$u = v = x_1$ ou $u = v = x_2$ ou $u = x_1, v = x_2$ ou $u = x_2, v = x_1$

- Pour les 2 premiers cas, on a $u = v$, donc
  u est égal à une des solutions u_1, u_2, u_3, u_4 de l'équation bicarrée :
  $P(x, x) = \alpha_1 x^4 + (2\beta_1 + \gamma_1)x^2 + \delta_1$ et
  les solutions des équations :
  $R(u_j, x) = 0$ pour $j = 1, 2, 3, 4$
  donneront les valeurs de w.
  Si on cherche seulement des solutions réelles en u, v, w il faudra faire des vérifications...
- Pour les 2 derniers cas, on a $u = x_1$ et $v = x_2$ ou $u = x_2$ et $v = x_1$ donc :
  $uv = x_1 x_2 = c/a$ et $u + v = x_1 + x_2 = -b/a$ donc
  $u^2 v^2 = x_1^2 x_2^2 = c^2/a^2$ et
  $u^2 + v^2 = (u + v)^2 - 2uv = (x_1 + x_2)^2 - 2x_1 x_2 = (b^2 - 2ca)/a^2$.
  On doit donc résoudre l'équation bicarrée en w $Eq = P(x_1, x_2) = 0$ que l'on écrit en fonction de a, b, c :
  $Eq = P(x_1, x_2) = \alpha_1 c^2 + \beta_1(b^2 - 2ca) + \gamma_1 ca + \delta_1 a^2 = 0$
  Avec `Xcas` , il faudra taper pour résoudre $Eq = 0$:

```
a:=alpha*w^2+beta;
b:=gamma*w;
c:=eta*w^2+delta;
Eq:=normal(alpha1*c^2+beta1*(b^2-2*c*a)+gamma1*c*a+
    delta1*a^2);
solve(Eq=0,w);
```

Cas P, Q et R sont symétriques et ont les mêmes coefficients
On a : $\alpha_1 = \alpha$, $\beta_1 = \beta = \eta$, $\gamma_1 = \gamma$, $\delta_1 = \delta$
Il faut résoudre $Eq = 0$ qui est une équation bicarrée en w, mais dans ce cas Eq se factorise.
On tape :

```
a:=alpha*w^2+beta;
b:=gamma*w;
c:=beta*w^2+delta;
Eq:=normal(alpha*c^2+beta*(b^2-2*c*a)+gamma*c*a+
    delta*a^2);
```

On factorise Eq, on tape : `factor(Eq)`
On obtient :
`(alpha*delta-beta^2+beta*gamma)*`
`(alpha*w^4+delta+2*w^2*beta+w^2*gamma)`
On voit que `(alpha*delta-beta^2+beta*gamma)` est un facteur de `Eq` donc :

- si `(alpha*delta-beta^2+beta*gamma)=0` (on verra dans l'étude qui suit que c'est le cas pour le cyclohexane), alors le polynôme Eq est identique à 0 et pour toutes les valeurs de w, si x_1 et x_2 sont les solutions différentes de $R(w, x) = (\alpha w^2 + \beta)x^2 + (\gamma w)x + (\beta w^2 + \delta) = 0$ alors, $R(u, v) = 0, R(v, w) = 0, R(w, u) = 0$ a une infinité de solutions :
  $u = x_1$, $v = x_2$ et w où x_1 et x_2 sont les solutions différentes dépendant de w de l'équation en x, $R(x, w) = 0$.
  Si on cherche seulement des solutions réelles en u, v, w il faudra faire des vérifications...
- si `(alpha*delta-beta^2+beta*gamma)`$\neq$ `0`
  Il reste à résoudre :
  `alpha*w^4+(2*beta+gamma)*w^2+delta=0`
  ce qui donne en général 4 solutions pour w : $w_1, w_2, -w_1, -w_2$.
  Là aussi si on cherche seulement des solutions réelles en u, v, w il faudra faire des vérifications...

Cas P est symétrique et Q et R ont les mêmes coefficients
C'est le cas du glucose.
Il n'y a pas de simplification, il faut résoudre $Eq = 0$ et taper :

```
a:=alpha*w^2+beta;
b:=gamma*w;
c:=eta*w^2+delta;
Eq:=normal(alpha1*c^2+beta1*(b^2-2*c*a)+ gamma1*c*a+
    delta1*a^2);
solve(Eq=0,w);
```

15.6.3 Molécule du cyclohexane

On choisit ici la molécule du cyclohexane constituée de 6 atomes de carbone et dans la modélisation ci-dessous, on ne prend en compte que les liaisons entre les atomes constituant le cycle, et non celles avec les atomes voisins du cycle.
Pour la molécule du cyclohexane la chimie nous impose :



Le triangle ACE est équilatéral et les triangles ABC, CDE, EFA sont isocèles avec $AB = AC = DC = DE = FE = FA =$1 unité de longueur et les angles B, D, F de ces triangles ainsi que les angles $\widehat{FAB}$, $\widehat{BCD}$ et $\widehat{DEF}$ sont égaux à 109 degrés. Donc on a :
$l = AC = AE = CE = 2\cos(\pi/2 - 109\pi/360) = 2\cos(71\pi/360)$.
D'après les chimistes, il y a 2 conformations appelées "chaise" et "bateau" : les points B, D, F sont d'un même côté du plan ACE (c'est le "bateau") ou 2 des points B, D, F sont d'un même côté du plan ACE et l'autre point est de l'autre côté (c'est la "chaise").

La "chaise" et le "bateau" symétrique en géométrie

L'étude géométrique faite ici est simplifiée car on ne cherche que les solutions admettant des symétries mais on verra qu'il y beaucoup d'autres solutions.
On peut commencer par faire un modèle en carton du problème (les traits marquent les pliures).



et on améne en coincidence :
CDD avec CD en pliant selon BC et EC,
EFF avec EF en pliant selon ED et EA, et
ABB avec AB en pliant selon FA et AC.
Lorsqu'on plie selon AC, le point B se déplace dans le plan médiateur de AC sur le cercle de centre le milieu H de AC rayon $BH = \sin(a_1)$ où $a_1 = \widehat{BAC}$.
Lorsque FF est en F pour pouvoir amener DD en D et BB en B, il faut que $BD = BF$ donc B se trouve dans le plan médiateur de DF sur le cercle de centre le milieu H_1 de FD et de rayon $AC\sqrt{3}/2$.
Donc quand la molécule est formée, B est à l'intersection du cercle de centre E et de rayon $AC\sqrt{3}/2$ du plan médiateur de FD et du cercle de centre H et de rayon $BH = \sin(a_1)$ du plan médiateur de AC.
Quand la molécule est formée, cherchons tout d'abord les solutions lorsque FD est parallèle à AC i.e. lorsque le plan médiateur de AC et le plan médiateur de FD sont confondus. Cela entraine que les angles des plans CDE, EFA avec le plan ACE sont égaux si on suppose que F et D sont d'un même côté du plan ACE.
En effet les points B, F et D sont soit tous les trois d'un même côté du plan ACE (c'est la forme "bateau"), soit deux de ces points sont d'un même côté du plan ACE et le troisième point est de l'autre côté (c'est la forme "chaise").
Dans la suite, on suppose que F et D sont au dessus du plan ACE lorsque l'on se place dans le repère orthonormé directe $(A, \overrightarrow{i}, \overrightarrow{j}, \overrightarrow{k})$ tel que $\overrightarrow{i}$ dirige AC et $\overrightarrow{j}$ soit dans le plan ACE pour que l'ordonnée de E soit positive.
Du fait des symétries, on ne va considérer que les 2 possibilités :
soit B, F et D sont soit tous les trois au dessus du plan ACE,
soit F et D sont au dessus du plan ACE et B est en dessous de ce plan.

— B, F et D sont soit tous les trois au dessus du plan ACE
  Cherchons les coordonnées de B lorsque les angles des plans ABC, CDE, EFA avec le plan ACE sont égaux. Dans ce cas le triangle BDF est un

triangle équilatéral et du fait de la symétrie du cycle, si on regarde la vue de dessus, on voit que BDF forme avec ACE une étoile à 6 branches.
La vue de dessus :



les 2 plans ACE et BDF sont parallèles et distant de $z_B = h$.
Donc si on pose :
`b:=109/180*pi; a1:=normal(pi/2-b/2);AC=l:=2*cos(a1)`
les coordonnées de B sont :
$x_B = \cos(a_1), y_B = -\frac{l*\sqrt{3}}{6}, z_B = \sqrt{\sin(a_1)^2 - \frac{l^2}{12}} = \sqrt{\sin(a_1)^2 - \frac{\cos(a_1)^2}{3}}$
Les coordonnées de D sont :
$x_D = l, y_D = \frac{l*\sqrt{3}}{3}, h = z_D = z_B = \sqrt{\sin(a_1)^2 - \frac{\cos(a_1)^2}{3}}$
les coordonnées de F sont :
$x_F = 0, y_F = \frac{l*\sqrt{3}}{3}, z_F = z_B = \sqrt{\sin(a_1)^2 - \frac{\cos(a_1)^2}{3}}$
Vue en 3d du bateau symétrique
Avec `Xcas` on ouvre un niveau de géométrie $3d$ et on tape :

```
a1:=71/360*pi;
l:=2*cos(a1);
h:=sqrt(sin(a1)^2-cos(a1)^2/3);
triangle(point(0,0,0),point(l,0,0),
         point(l/2,l*sqrt(3)/2,0),affichage=1);
triangle(point(0,0,0),point(l,0,0),
         point(cos(a1),-l*sqrt(3)/6,h));
triangle(point(l,0,0),point(l/2,l*sqrt(3)/2,0),
         point(l,l*sqrt(3)/3,h));
triangle(point(0,0,0),point(l/2,l*sqrt(3)/2,0),
         point(0,l*sqrt(3)/3,h));
```

```
triangle(point(cos(a1),-l*sqrt(3)/6,h),point(l,l*sqrt(3)/3,h),
          point(0,l*sqrt(3)/3,h),affichage=4);
```

On obtient (en rouge le triangle ACE et en bleu le triangle BDF) :



— F et D sont au dessus du plan ACE et B est en dessous de ce plan.
Cherchons les coordonnées de B lorsque les angles des plans CDE, EFA avec le plan ACE sont égaux et lorsque B a une cote négative.
Faisons une coupe selon le plan médiateur de AC (qui est aussi le plan médiateur de FD).
Ce plan médiateur est parallèle à $A\overrightarrow{j}\overrightarrow{k}$ et on munit ce plan médiateur du repère $H\overrightarrow{j}\overrightarrow{k}$ (H milieu de AC se projéte en A).
H_1 le milieu de FD a pour coordonnées :
$2\cos(a_1)/\sqrt(3), h = \sqrt{\sin(a_1)^2 - \cos(a_1)^2/3}$
Traçons le cercle de centre H et de rayon $l * \sqrt(3)/6$ et le cercle de centre H_1 et de rayon $\cos(a_1)$. Les 2 points d'intersections correspondent à deux points B qui sont solutions : on retrouve ainsi la solution précédente ainsi qu'une autre solution.
Avec `Xcas`, on tape :

```
a1:=71/360*pi;
l:=2*cos(a1);
H:=point(0);
H1:=point(2*cos(a1)/sqrt(3),sqrt(sin(a1)^2-cos(a1)^2/3));
CH1:=cercle(H1,l*sqrt(3)/2):;CH1;
CH:=cercle(H,sin(a1)):;CH;
S:=inter(CH,CH1):;
B0:=S[0];
B1:=S[1];
```

On a :

`evalf(coordonnees(B0))` renvoie `[-0.470029813674,0.341010112795]`
`evalf(coordonnees(B1))` renvoie `[-0.142077142237,-0.563054178943]`
Ce qui donne comme coordonnèes des points B qui sont les solutions :
`[0.814115518356,-0.470029813674,0.341010112795]` et
`[0.814115518356,-0.142077142237,-0.563054178943]`
puisque `evalf(cos(a1))` renvoie `0.814115518356`
et on obtient :



Le point `SB0:=point(cos(a1),l*sqrt(3)/6,h)` ($h := \sqrt{\sin(a1)^2 - \cos(a1)^2/3}$) est le point `B` trouvé précédemment et correspond à `B0` du plan médiateur de AC et la deuxième solution $SB1$ est le point `B1` du plan médiateur de AC.
$B1$ est donc le symétrique de $B0(\cos(a_1)\sqrt{3}/3, \sqrt{\sin(a_1)^2 - \cos(a_1)^2/3})$ par rapport à la droite $HH1$ d'équation :
$y = hx/(2 * \cos(a_1)\sqrt{3})$.
On tape :

```
LB1:=simplify(coordonnees(symetrie(droite(y=h1*x/(2*k1)),
point(k1,h1))))
```

On obtient :

```
[(5*h1^2*k1-4*k1^3)/(h1^2+4*k1^2),
(h1^3-8*h1*k1^2)/(h1^2+4*k1^2)]
```

On tape :

```
h1:=sqrt(sin(a1)^2-cos(a1)^2/3);
k1:=cos(a1)/sqrt(3);
trigsin(tlin(LB1))
```

On obtient les coordonnées de B_1 :

```
[sqrt(3)*(2*cos(49*pi/120)-cos(71*pi/360))/3,
((-2*sqrt(12*sin(71*pi/360)^2-3))*cos(71*pi/180)-
(sqrt(12*sin(71*pi/360)^2-3)))/3]
```

après un `evalf` on trouve les coordonnées approchées de B_1 :
`[-0.142077142237,-0.563054178943]`
La deuxième solution $SB1$ a donc comme coordonnées :

```
[cos(a),sqrt(3)*(2*cos(49*pi/120)-cos(71*pi/360))/3,
((-2*sqrt(12*sin(71*pi/360)^2-3))*cos(71*pi/180)-
(sqrt(12*sin(71*pi/360)^2-3)))/3]])
```

ou encore en fonction de h et l :
$[(15*\sqrt{3}*h^2*l+(-(\sqrt{3}))*l^3)/(18*h^2+6*l^2),$
$(3*h^3-2*h*l^2)/(3*h^2+l^2)]$

Vue en 3d de la chaise symétrique

Avec `Xcas` on ouvre un niveau de géométrie $3d$ et on tape :

```
b:=109/180*pi;
a1:=normal(pi/2-b/2);
l:=2*cos(a);
h:=sqrt(sin(a1)^2-cos(a1)^2/3);
B1:=point(cos(a),sqrt(3)*(2*cos(49*pi/120)-cos(71*pi/360))/3,
    ((-2*sqrt(12*sin(71*pi/360)^2-3))*cos(71*pi/180)-
     (sqrt(12*sin(71*pi/360)^2-3)))/3):;
triangle(point(0,0,0),point(l,0,0),point(l/2,l*sqrt(3)/2,0),
         affichage=1);
triangle(point(0,0,0),point(l,0,0),B1);
triangle(point(l,0,0),point(l/2,l*sqrt(3)/2,0),
         point(l,l*sqrt(3)/3,h));
triangle(point(0,0,0),point(l/2,l*sqrt(3)/2,0),
         point(0,l*sqrt(3)/3,h));
segment(point(l,l*sqrt(3)/3,h),point(0,l*sqrt(3)/3,h),
        affichage=4);
segment(point(l,l*sqrt(3)/3,h),B1,affichage=2);
segment(B1,point(0,l*sqrt(3)/3,h),affichage=2);
```

On obtient (en rouge le triangle ACE, en bleu le segment FD et en vert les segments DB et FB) :



Solution analytique de la forme du cyclohexane

Voilà ce qu suggère le texte de l'épreuve de modélisation de l'agrégation pour trouver les coordonnées de B, D et F.
On se place dans le repère orthonormé directe $(A, \overrightarrow{i}, \overrightarrow{j}, \overrightarrow{k})$ tel que $\overrightarrow{i}$ dirige AC et $\overrightarrow{j}$ soit dans le plan ACE (ordonnée de E est positive).
Comme B se déplace sur le cercle de centre H (milieu de AC) et de rayon BH dans le plan $x = AC/2$, on peut paramétrer ce cercle sous la forme :
$x = AC/2, y = BH\dfrac{1-u^2}{1+u^2}, z = BH\dfrac{2u}{1+u^2}$ ($u = \tan(t/2)$ lorsque t est l'angle du plan ABC avec le plan ACE).
On a ABC est isocèle de sommet B et l'angle A de ce triangle a comme mesure en radians : $a_1 = \pi/2 - 109 * \pi/360 = 71\pi/360$
Si $AB = 1$ on a $AC/2 = \cos(a_1)$ et $BH = \sin(a_1)$ donc les coordonnées de B sont :
$x = \cos(a_1)/2, y = \sin(a_1)\dfrac{1-u^2}{1+u^2}, z = \sin(a_1)\dfrac{2u}{1+u^2}$.
Pour avoir les coordonnées de F en fonction de v, il suffit d'écrire les coordonnées de F en imaginant AC et AE superposés, puis de faire une rotation appropriée.
Ici, F est l'image du point SB symétrique de B par rapport à AC (de paramètre v) par une rotation d'axe $A\overrightarrow{k}$ d'angle $a = \pi/3$.
Pour avoir les coordonnèes de D, on dit que D est l'image de SB (de paramètre w) par la rotation de centre $C\overrightarrow{k}$ et d'angle $-\pi/3$.
On évalue $sa1 = \sin(a_1)$, $ca1 = \cos(a_1)$ et $cb = \cos(\pi - 2a_1)$ en fonction de :
$t = \tan(a_1/2) = \tan(71\pi/720)$
On a :
$a_1 = 71\pi/360$
$t = \tan(a_1/2)$

$sa1 = \sin(71\pi/360) = 2t/(1+t^2)$
$ca1 = \cos(71\pi/360) = (1-t^2)/(1+t^2)$
$cb := \cos(109\pi/180) = \cos(\pi - 2a_1) = -(2ca1^2 - 1)$
$a_1 = 71\pi/360.$
On tape :

```
//t:=tan(71*pi/720);
a:=pi/3;
ca1:=(1-t^2)/(1+t^2); sa1:=2*t/(1+t^2);
cb:=-(2*ca1^2-1);
cA:=coordinates(point(0,0,0));
cC:=coordinates(point(2*ca1,0,0));
cB:=normal(coordinates(point(ca1,sa1*(1-u^2)/(1+u^2),
           sa1*2*u/(1+u^2))));
cE:=normal(coordinates(point(ca1,ca1*sqrt(3),0)));
cF:=normal(coordinates(rotation(line(x=0,y=0),a,
    point(ca1,-sa1*(1-v^2)/(1+v^2),sa1*2*v/(1+v^2)))));
cD:=normal(coordinates(rotation(line(x=2*ca1,y=0),-a,
    point(ca1,-sa1*(1-w^2)/(1+w^2),sa1*2*w/(1+w^2)))));
P:=numer(cB*cF-cb);
Q:=numer((cD-cE)*(cF-cE)-cb);
R:=numer((cB-cC)*(cD-cC)-cb);
```

On tape :
`P:=normal(P)`
`Q:=normal(Q)`
`R:=normal(R)`
On obtient pour R :

```
3/2*t^4*u^2*w^2+3/2*t^4*u^2+3/2*t^4*w^2+
3/2*t^4+2*sqrt(3)*t^3*u^2*w^2+(-2*sqrt(3))*t^3-
9*t^2*u^2*w^2-5*t^2*u^2+16*t^2*u*w-5*t^2*w^2-
9*t^2+(-2*sqrt(3))*t*u^2*w^2+2*sqrt(3)*t+
3/2*u^2*w^2+3/2*u^2+3/2*w^2+3/2
```

On sait qu'en raison des symétries de la configuration les conditions $P = 0, Q = 0, R = 0$ peuvent s'exprimer en fonction de R.
Pour le vérifier, on tape :
`simplify(R-subst(P,v,w))`
On obtient : `0`
et on tape :
`simplify(R-subst(Q,v,u))`
On obtient : `0`
On définit les coefficients α, β, γ, δ, pour que :
$R = \alpha u^2w^2 + \beta(u^2 + w^2) + \gamma uw + \delta$
On tape :

```
LC:=coeffs(R,u)
alpha,x,beta:=coeffs(LC[0],w);
gamma,x:=coeffs(LC[1],w);
```

```
beta,x,delta:=coeffs(LC[2],w);
purge(x);
```

Pour avoir une valeur approchée de R, on tape :
`a1:=71*pi/360.;t:=tan(a1/2);`
`normal(R)`
On obtient :

```
-0.803377348547*u^2*w^2+2.00684135654*u^2+3.27890002175*u*w+
2.00684135654*w^2+3.17761005070
```

Application du lemme pour la molécule de cyclohexane

On cherche les solutions approchées telles que $u = v$.
On tape pour avoir les solutions telles que $u = w$:
```
purge(t,a1);
S1:=solve(subst(P,v,u)=0,u)
t:=tan(71*pi/720);
```
On obtient :

```
[(sqrt(-9*t^4+30*t^2-9))/(3*t^2+(-2*sqrt(3))*t-3),
-(sqrt(-9*t^4+30*t^2-9))/(3*t^2+(-2*sqrt(3))*t-3),
(sqrt(-t^4+14*t^2-1))/(t^2+2*sqrt(3)*t-1),
-(sqrt(-t^4+14*t^2-1))/(t^2+2*sqrt(3)*t-1)]
```

On tape `t:=tan(71*pi/720) evalf(S1)`
On obtient :
```
[-0.645454288199*i,0.645454288199*i,
3.08123639142,-3.08123639142]
```
On tape pour avoir les solutions u, v, w telles que $u = v = 3.08123639142$:
`solve(subst(R,u,3.08123639142)=0,w)`
On obtient 2 valeurs de w :
`[-1.28367770808,3.08123639141]`
Donc les solutions qui vérifient $u = w$ sont :
$u, v, w = 3.08123639141, 3.08123639141, 3.08123639141$
$u, v, w = 3.08123639141, 3.08123639141, -1.28367770808$
ainsi que les solutions symétriques :
$u, v, w = 3.08123639141, -1.28367770808, 3.08123639141$
$u, v, w = -1.28367770808, 3.08123639141, 3.08123639141$
ainsi que les solutions opposées :
$u, v, w = -3.08123639141, -3.08123639141, -3.08123639141$
$u, v, w = -3.08123639141, -3.08123639141, 1.28367770808$
$u, v, w = -3.08123639141, 1.28367770808, -3.08123639141$
$u, v, w = 1.28367770808, -3.08123639141, -3.08123639141$
On cherche les solutions approchées telles que $u \neq v$.
On tape :
`normal(beta^2-beta*gamma-alpha*delta)`
On obtient : `0`
Donc lorsque l'on fixe w les solutions différentes u_1, u_2 (dépendant de w) de l'équation en u `R=0` sont solution de : `subst(subst(P,u,u1),v,u2)=0`.

On tape :
`S2:=normal(evalf(solve(R=0,u)))`
On obtient :
`[(1.63945001088*w-sqrt(d))/(0.803377348547*w^2-2.00684135654),`
`(1.63945001088*w+sqrt(d))/(0.803377348547*w^2-2.00684135654)]`
avec : `d:=1.61225088797*w^4+1.21320404513*w^2-6.37695926481`
On cherche le signe de d, on tape :
`factor(d)`
On obtient :

```
1.61225088797*(w-1.28367770808)*(w+1.28367770808)*
(w^2+2.40031931071)}
```

Donc $d \geq 0$ si $w \geq 1.28367770808$ ou si $w \leq -1.28367770808$.
On a alors des solutions réelles u, v qui sont x_1, x_2 les solutions de `solve(R=0,u)` :
$\dfrac{1.63945001088 * w - \sqrt{d}}{0.803377348547 * w^2 - 2.00684135654}, \dfrac{1.63945001088 * w + \sqrt{d}}{0.803377348547 * w^2 - 2.00684135654}$
avec $d = 1.61225088797 * w^4 + 1.21320404513 * w^2 - 6.37695926481$ et si $w = 1.28367770808$ ou si $w = -1.28367770808$ on a $d = 0$ donc $u = v$ et on retouve une des solutions trouvées précédemment à savoir :
$u, v, w = 3.08123639148, 3.08123639148, -1.28367770808$ et
$u, v, w = -3.08123639148, -3.08123639148, 1.28367770808$.
On remarque que si $w^2 > 1.28367770808^2 = 1.6478284582$ alors les solutions de l'équation en u `R=0`, seront de même signe si :
`normal(subst(R,u,0))=2.00684135654*w^2+3.17761005075` est de même signe que :
`normal(coeffs(R,u)[0])=-0.803377348547*w^2+2.00684135654`
c'est à dire si $-0.803377348547 * w^2 + 2.00684135654 > 0$
c'est à dire si $w^2 < 2.49800589993$. On a $\sqrt{2.49800589993} = 1.58050811448$ donc :
Les solutions u, v qui dépendent de w sont de même signe négatif si :
$-1.58050811448 < w < -1.28367770808$ ou $1.28367770808 < w < 1.58050811448$.
et sinon sont de signe opposé.

Des exemples de solutions

On tape :

```
d(w):=1.61225088797*w^4+1.21320404513*w^2-6.37695926481
u(w):=(1.63945001088*w-sqrt(d(w)))/(0.803377348547*w^2-2.00684135654)
v(w):=(1.63945001088*w+sqrt(d(w)))/(0.803377348547*w^2-2.00684135654)
u(1.5),v(1.5),1.5
```

On obtient :
`-1.67823888288,-23.0070286629,1.5`
On tape :
`u(2),v(2),2`
On obtient :
`-1.36553633921,6.80017108457,2`
On tape :
`u(3),v(3),3`
On obtient :
`-1.28387158972,3.16701448356,3`

On tape :
`u(30),v(30),30`
On obtient :
`-1.51735311476,1.65377825995,30`
Lorsque w tend vers l'infini (ce qui correspond à un angle plat entre le plan ACE et le plan CDE on tape :
On tape :
```
purge(w)
limit(u(w),w=inf)
```
On obtient :
`-1.58050811448`
On tape :
`limit(v(w),w=inf)`
On obtient :
`1.58050811448`
Remarque si $w = 3.08123639148$ alors $\sqrt{d} = 12.266363249$, on trouve alors $u = -1.28367770808$ et $v = 3.08123639148$ car si $w > 1.28367770808$ ou si $w < -1.28367770808$, u et v seront différents.
Pour retrouver la solution $u = v = w$ et faudrait résoudre les équations en w :
`subst(subst(P,u,x1),v,x1)=0` et `subst(subst(P,u,x2),v,x2)=0`
Mais il est plus simple de résoudre directement l'équation en u :
`subst(P,v,u)=0` qui a 4 solutions u_j ($j = 1..4$) puis de résoudre pour chaque solution l'équation en w :
`subst(R,u,uj)=0`.

Avec le résultant

On tape :
```
purge(t);
P:=3/2*t^4*u^2*v^2+3/2*t^4*u^2+3/2*t^4*v^2+
3/2*t^4+2*sqrt(3)*t^3*u^2*v^2+(-2*sqrt(3))*t^3-
9*t^2*u^2*v^2-5*t^2*u^2+16*t^2*u*v-5*t^2*v^2-
9*t^2+(-2*sqrt(3))*t*u^2*v^2+2*sqrt(3)*t+
3/2*u^2*v^2+3/2*u^2+3/2*v^2+3/2;

Q:=3/2*t^4*v^2*w^2+3/2*t^4*v^2+3/2*t^4*w^2+
3/2*t^4+2*sqrt(3)*t^3*v^2*w^2+(-2*sqrt(3))*t^3-
9*t^2*v^2*w^2-5*t^2*v^2+16*t^2*v*w-5*t^2*w^2-
9*t^2+(-2*sqrt(3))*t*v^2*w^2+2*sqrt(3)*t+
3/2*v^2*w^2+3/2*v^2+3/2*w^2+3/2;

R:=3/2*t^4*u^2*w^2+3/2*t^4*u^2+3/2*t^4*w^2+
3/2*t^4+2*sqrt(3)*t^3*u^2*w^2+(-2*sqrt(3))*t^3-
9*t^2*u^2*w^2-5*t^2*u^2+16*t^2*u*w-5*t^2*w^2-
9*t^2+(-2*sqrt(3))*t*u^2*w^2+2*sqrt(3)*t+
3/2*u^2*w^2+3/2*u^2+3/2*w^2+3/2
```
On tape pour éliminer w des équations $Q = 0, R = 0$:
`RQ:=resultant(Q,R,w)`

On factorise RQ, on tape :
`fRQ:=factor(RQ)`
On obtient :
`1024*t^4*(t-(sqrt(3)))*(t+(sqrt(3))/3)*(t+sqrt(3))*(3*t-(sqrt(3)))*(v-`
On trouve $u = v$ ou $fRQ[8] = 0$
On tape pour éliminer v des équations $P = 0, fRQ[8] = 0$:
`PfRQ:=resultant(P,fRQ[8],v)`
On obtient :
`0`
On tape :
`normal(2*P-fRQ[8])`
On obtient :
`0`
donc $2P$ et $fRQ[8]$ sont les mêmes polynômes.
On sait qu'il existe U et V tel que :
$U * Q + V * R = RQ$ (on a :`U,V:=abcuv(Q,R,RQ,w)`) On a montré que :
$RQ = U * Q + V * R = k * (u - v) * P$ où k est une constante.
On fixe w. Si u et v vérifient $Q = 0$ et $R = 0$ alors $(u - v) * P = 0$ c'est à dire :
si u et v vérifient $Q = 0$ et $R = 0$ alors soit $u = v$ soit $P = 0$.
Donc on retrouve les solutions :
$(u, v = u, w$ lorsque $u = u_j$ avec u_j solution de `solve(subst(P,v,u))=0,u)`
et $w = w_k$ avec w_k solution de `solve(subst(R,u,uj))=0,w)` et
$u, v \neq u, w$ lorsque u et v sont les solutions différentes dépendant de w de `solve(R=0,u)`
Il reste a regarder quand ces solutions sont réelles.

15.6.4 La molécule de glucose

On choisit ici la molécule de glucose qui est constituée de 6 atomes (un atome d'oxygène et 5 atomes de carbone) et dans la modélisation ci-dessous, nous ne prenons en compte que les liaisons entre les atomes constituant le cycle, et non celles avec les atomes voisins du cycle.
Pour la molécule de glucose la chimie nous impose :



$L_1 = AB = AF = 1.43$ Angström,
$L_2 = BC = CD = DE = EF = 1.54$ Angström
$t_1 = \widehat{BAF} = 106 * \pi/180$ radians

$t_2 = \widehat{ABC} = \widehat{BCD} = \widehat{CDE} = \widehat{DEF} = \widehat{EFA} = 109 * \pi/180$ radians
On calcule :
$l_1 = AC = AE = \sqrt{AB^2 + BC^2 - 2AB * BC * \cos(109\pi/180)}$
$l_1 = \sqrt{L_1^2 + L_2^2 - 2L_1L_2\cos(t_2)}$
$l_2 = CE = \sqrt{2CD^2 - 2CD^2 * \cos(109\pi/180)} = \sqrt{2L_2^2 - 2L_2^2\cos(t_2)}$
$l_3 = AH = \sqrt{l1^2 - l2^2/4}$ (où H est le milieu de CE)
$l_4 = FB = \sqrt{2L_1^2 - 2L_1^2\cos(t_1)}$ (l_4 est le troisième côté du triangle isocèle d'angle 106 degrés et de côtés 1.43).
$h_4 = DJ = \sqrt{l_2^2 - l_4^2/4}$ (où J est le milieu de BF) On pose :
$c_0 = \widehat{DCE}\pi/2 - t_2/2 = 71 * \pi/360$.
$c_1 = \widehat{ECA}$
$c_2 = \widehat{ACB}$
$a_3 = \widehat{CAB}$
$cc1 = \cos(c_1) = l_2/(2l_1)$
$sc1 = \sin(c_1) = l_3/l_1$
$sc2 = \sin(c2) = L1 * \sin(t_2)/l_1$
$cc2 = \cos(c2) = (l_1^2 + L_2^2 - L_1^2)/(2 * l_1 * L_2) = (L_2 - L_1 * \cos(t_2))/l_1;$
$sa3 = \sin(a3) = L2 * \sin(t_2)/l_1$
$ca3 = \cos(a3) = (L_1^2 + l_1^2 - L_2^2)/(2 * L_1 * l_1) = (L_1 - L_2 * \cos(t_2)/l_1$
On tape pour un calcul approché de AC et AE :
`l1:=sqrt(1.43^2+1.54^2-2*1.43*1.54*cos(109*pi/180))`
On obtient : `AC=AE==2.4187667063`
On tape pour un calcul approché de CE :
`l2:=sqrt(2*1.54^2-2*1.54^2*cos(109*pi/180));`
On obtient : `2.50747579654`
On tape pour un calcul approché de AH :
`l3:=sqrt(l1^2-l2^2/4);`
On obtient : `2.50747579654`
On tape pour un calcul approché de $l_4 = FB$ (lorsque la molécule est formée) :
`l4:=sqrt(2*1.43^2-2*1.43^2*cos(106*pi/180));`
On obtient : `2.28409755874`

La "chaise" et le "bateau" symétrique du glucose en géométrie

Le modéle en carton

On peut commencer par faire un modéle en carton du problème : parmi les triangles qui bordent ACE il y a 3 triangles isocèles ($CBDD$, CDE et $EDFF$) d'angle 109 degrés et de côtés 5 cm, 1 triangle isocèle ($AFBB$) d'angle 106 degrés et de côtés 5*1.43/1.54 cm$\simeq$ 4.64 cm et 2 triangles (ABC et AFE) de cotés 5 cm et 4.64 cm et d'angle 109 degrés.
Le modéle :



Puis, on forme la molécule en amenant en coincidence :
CDD avec CD en pliant selon BC et EC, EFF avec EF en pliant selon ED et EA, et ABB avec AB en pliant selon FA et AC.

Les solutions symétriques en géométrie

Pour privilégier, la sym'etrie par rapport à la médiatrice de CE, il semble plus judicieux de se placer dans le repère orthonormé direct :
$C, \overrightarrow{i}, \overrightarrow{j}, \overrightarrow{k}$ tel que $\overrightarrow{i}$ dirige $\overrightarrow{CE}$, $\overrightarrow{j}$ dirige $\overrightarrow{HA}$ où H est le milieu de CE.

Voici les notations :



On tape :

```
L1:=1.43;L2:=1.54;
t1:=106*pi/180.;
t2:=109*pi/180.;
```

```
c0:=71*pi/360.;
l1:=sqrt(normal(L1^2+L2^2-2*L1*L2*cos(t2)));
l2:=sqrt(normal(2*L2^2-2*L2^2*cos(t2)));
l3:=sqrt(normal(l1^2-l2^2/4));
l4:=sqrt(normal(2*L1^2-2*L1^2*cos(t1)));
h4:=sqrt(normal(l2^2-l4^2/4));
sc1:=sin(c1);
cc1:=cos(c1);
sc2:=sin(c2);
cc2:=cos(c2);
sa3:=sin(a3);
ca3:=cos(a3);
```

Le but est de calculer x_0, y_0, z_0 les coordonnées de B lorsque la molécule est formée de façon à ce que B et D soient symétriques par rapport au plan médiateur de CF et on suppose $z_0 > 0$.
Pour cela nous projetons la molécule de glucose 3d symétrique sur le plan ACE et sur le plan médiateur de CE.

Faisons maintenant une projection sur le plan ACE



B se projette en N sur AC et on a $AN = L1\cos(a_3)$
B se projette en B_p sur le plan ACE et B_pN est perpendiculaire à AC (c'est le théorème des 3 perpendiculaires)
B_p se projette en M_p sur la médiatrice de CE et on a $B_pM_p = l4/2$.
Si K est l'intersection de B_pN et de la médiatrice de CE, on a :

$$\frac{AN}{AK} = \sin(c_1) \text{ et } AN = L_1\cos(a_3) \text{ donc } AK = L_1\cos(a_3)/\sin(c_1)$$

$$\frac{B_pM_p}{KM_p} = \tan(c_1) \text{ donc } KM_p = l_4/(2\tan(c_1)) = l_4 * \cos(c_1)/(2*\sin(c_1)) \text{ et}$$

$$AM_p = AK - KM_p = L1\cos(a_3)/\sin(c_1) - l_4 * \cos(c_1)/(2*\sin(c_1))$$

Donc B_p a pour coordonnées :
$x_0 = l_2/2 - l_4/2,\ y_0 = l_3 - L_1\cos(a_3)/\sin(c_1) + l_4/(2\tan(c_1))$.
Avec les notations que l'on a choisies on tape :
`Bp:=point(l2/2-l4/2,l3-L1*ca3/sc1+l4*cc1/(2*sc1))`
Cherchons la cote z_0 de B.
On sait que :
B se projette en N sur AC
$AB = L1$
$BN = L1\sin(a_3)$ donc
$z_0^2 = AB^2 - AB_p^2 = L_1^2 - l_4^2/4 - AM_p^2 =$
$z_0^2 = L_1^2 - l_4^2/4 - (L1\cos(a_3)/\sin(c_1) + l_4\cos(c_1)/(2\sin(c_1)))^2$
Avec les notations que l'on a choisies on tape :
`z0:=normal(sqrt(L1^2-longueur2(A,Bp)));`
Donc B a pour coordonnées $x0, y0, z0$ soit :
$0.111689118901,\ 1.42546684563,\ 0.571987598417$
et F a pour coordonnées $x0 + l4, y0, z0$:
$2.39578667764,\ 1.42546684563,\ 0.571987598417$

Faisons maintenant une coupe selon le plan médiateur de CE
Soit M la projection de B sur le plan médiateur de CE
Lorsque la molécule symétrique est formée, le point D se trouve dans ce plan médiateur sur le cercle de centre H et de rayon $L_2\sin(c_0)$.
Le triangle BDF est un triangle isocèle de côtés l_4, l_2, l_2 et de hauteur :
$h_4 = \sqrt{l_2^2 - l_4^2/4}$ donc
D est aussi sur le cercle de centre M et de rayon h_4.
On fait la figure dans le plan médiateur avec comme repère Hyz.



On tape :

```
H:=point(0);
```

```
A:=point(l3);
M:=point(y0,z0);
h4:=sqrt(l2^2-l4^2/4);
C1:=cercle(M,h4);
C2:=cercle(H,L2*sin(c0));
D0,D1:=inter(CE1,CE2);
coordonnees(D0);
coordonnees(D1);
```

On obtient pour D_0 :
`[-0.302153034804,-0.84169164544]`
On obtient pour D_1 :
`[-0.800148906999,0.399378278168]`
Donc dans l'espace le point D a comme abscisse $l_2/2 = 1.25373789827$ et a donc comme coordonnèes 3d :
$[1.25373789827, -0.302153034804, -0.84169164544]$
ou :
$[1.25373789827, -0.800148906999, 0.399378278168]$
Il y a donc comme solutions :
B de coordonnées $0.111689118901,\ 1.42546684563,\ 0.571987598417$
F de coordonnées $2.39578667764,\ 1.42546684563,\ 0.571987598417$
D de coordonnées $1.25373789827,\ -0.302153034804,\ -0.84169164544$
soit
B de coordonnées $0.111689118901,\ 1.42546684563,\ 0.571987598417$
F de coordonnées $2.39578667764,\ 1.42546684563,\ 0.571987598417$
D de coordonnées $1.25373789827,\ -0.800148906999,\ 0.399378278168$
soit
B de coordonnées $0.111689118901,\ 1.42546684563,\ -0.571987598417$
F de coordonnées $2.39578667764,\ 1.42546684563,\ -0.571987598417$
D de coordonnées $1.25373789827,\ -0.302153034804,\ +0.84169164544$
soit
B de coordonnées $0.111689118901,\ 1.42546684563,\ -0.571987598417$
F de coordonnées $2.39578667764,\ 1.42546684563,\ -0.571987598417$
D de coordonnées $1.25373789827,\ -0.800148906999,\ -0.399378278168$

Solution analytique et coordonnées des points en 3d

Comme pour le cyclohexane on va paramétrer les points D, B, F.
On a : $CD = L_2$ et $\widehat{DCE} = c_0$ donc :
D est le point de coordonnées :
$L_2\cos(c_0), L_2\sin(c_0) * (1-w^2)/(1+w^2), L_2\sin(c_0) * 2 * w/(1+w^2)$
B est le transfomé du point de coordonnées :
$L_2\cos(c_2), -L_2\sin(c_2)*(1-u^2)/(1+u^2), 2L_2\sin(c_2)u/(1+u^2)$ dans la rotation d'axe $C\overrightarrow{k}$ et d'angle c_1 : F est le transfomé du point de coordonnées :
$L_2\cos(c_2), l_2 - L_2\sin(c_2) * (1-v^2)/(1+v^2), 2L_2\sin(c_2)v/(1+v^2)$ dans la rotation d'axe $E\overrightarrow{k}$ et d'angle $-c_1$.
On cherche à résoudre le système constitué des 3 relations :
$P = \overrightarrow{AB} * \overrightarrow{AF} - L_1^2 * \cos(t_1) = 0$

$Q = \overrightarrow{ED} * \overrightarrow{EF} - L_2^2 * \cos(t_2) = 0$
$R = \overrightarrow{CD} * \overrightarrow{CB} - L_2^2 * \cos(t_2) = 0$
Du fait de la symétrie par rapport à la médiatrice de CE, les polynômes Q et R auront les mêmes coefficients et le polynôme P sera symétrique en u et v.
Pour privilégier, la symétrie par rapport à la médiatrice de CE, on se place dans le repère orthonormé direct :
C, $\overrightarrow{i}$, $\overrightarrow{j}$, $\overrightarrow{k}$ tel que $\overrightarrow{i}$ dirige $\overrightarrow{CE}$, $\overrightarrow{j}$ dirige $\overrightarrow{HA}$ où H est le milieu de CE.
On ne fera pour le glucose que le calcul approché des coefficients de P, Q et R.
On a les mêmes notations qu'en géométrie :
$L_1 = 1.43; L_2 = 1.54; t_1 = 106 * pi/180.; t_2 = 109 * pi/180.; c_0 = 71 * pi/360.$
$c_1 = \widehat{ECA}$
$cc1 = \cos(c_1) = l_2/(2l_1)$
$sc1 = \sin(c_1) = l_3/l_1$
E est le point de coordonnées $(l_2, 0, 0)$
A a pour coordonnées : $l_2/2, l_3, 0$
$CD = L_2$ donc, D est le point de coordonnées :
$L_2 * \cos(c_0), L_2 * \sin(c_0) * (1 - w^2)/(1 + w^2), L_2 * \sin(c_0) * 2 * w/(1 + w^2)$
B est l'image du point de coordonnées :
$L_2 * \cos(c_2), -L_2 * \sin(c_2) * (1 - u^2)/(1 + u^2), L_2 * \sin(c_2) * 2 * u/(1 + u^2)]$
par la rotation d'axe $C\overrightarrow{k}$ d'angle c_1
F est l'image du point de coordonnées :
$[L2 * \cos(c_2), l_2 - L_2 * \sin(c_2) * (1 - v^2)/(1 + v^2), L_2 * \sin(c_2) * 2 * v/(1 + v^2)]$
par la rotation d'axe $E\overrightarrow{k}$ d'angle $-c_1$
F est aussi le symétrique par rapport au plan médiateur de CE du point B de paramètre w.
On tape :

```
L1:=1.43;L2:=1.54;
t1:=106*pi/180.;
t2:=109*pi/180.;
a2:=71*pi/360.;
l1:=sqrt(normal(L1^2+L2^2-2*L1*L2*cos(t2)));
l2:=sqrt(normal(2*L2^2-2*L2^2*cos(t2)));
l3:=sqrt(normal(l1^2-l2^2/4));
l4:=sqrt(normal(2*L1^2-2*L1^2*cos(t1)));
sc1:=sqrt(normal(l3^2/(l1^2)));
cc1:==sqrt(normal(l2^2/(4*l1^2)));
sc2:=L1/l1*sin(t2);
cc2:==(L2-L1*cos(t2))/l1;
cC:=coordonnees(point(0,0,0));
cE:=coordonnees(point(l2,0,0));
cA:=coordonnees(point(l2/2,l3,0));
cD:=normal([L2*cos(a2),L2*sin(a2)*(1-w^2)/(1+w^2),
            L2*sin(a2)*2*w/(1+w^2)]);
cB:=normal(coordonnees(rotation(droite(x=0,y=0),c1,
            point(L2*cc2,-L2*sc2*(1-u^2)/(1+u^2),
                  L2*sc2*2*u/(1+u^2)))));
cB:=normal(subst(subst(cB,cos(c1),cc1),sin(c1),sc1));
```

```
cF:=normal(subst([l2-cB[0],cB[1],cB[2]],u,v));
P:=numer((cB-cA)*(cF-cA)-L1^2*cos(t1));
Q:=numer((cD-cE)*(cF-cE)-L2^2*cos(t2));
R:=numer(cB*cD-L2^2*cos(t2));
```

On obtient les coefficients approchés de P, Q et de R.
$P := -0.918886458988 * u^2 * v^2 + 1.50972814948 * u^2 + 2.96432403045 * u * v + 1.50972814948 * v^2 + 2.56689196034$
$Q := -0.696642093056 * v^2 * w^2 + 2.05453942839 * v^2 + 3.07941052911 * v * w + 1.94741717452 * w^2 + 3.10242421803$
$R := -0.696642093056 * u^2 * w^2 + 2.05453942839 * u^2 + 3.07941052911 * u * w + 1.94741717452 * w^2 + 3.10242421803$
Pour faire un calcul exact, on pose :
`L1:=143/100;L2:=154/100;t1:=106*pi/180.;t2:=109*pi/180.`
Si $c_0 = \pi/2 - t_2/2 = 71 * \pi/360$, on évalue :
$\cos(t_2)$, $\cos(c_0)$ et $\sin(c_0)$ en fonction de $t := \tan(71\pi/720)$:
$\cos(t_2) = 2\cos(t_2/2)^2 - 1 = 2\sin(c_0)^2 - 1 = \dfrac{8t^2}{(1+t^2)^2} - 1 = \dfrac{-t^4+6t^2-1}{(1+t^2)^2}$
$\sin(t_2) = 2\cos(t_2/2) * \sin(t_2/2) = 2\sin(c_0)\cos(c_0) = \dfrac{4t(1-t^2)}{(1+t^2)^2} = \dfrac{4t-4t^3}{(1+t^2)^2}$

$$\cos(c_0) = \frac{1-t^2}{1+t^2}$$

$\sin(c_0) = \dfrac{2t}{1+t^2}$
On a donc :
$l_1 = AC = AE = \sqrt{L1^2 + L2^2 - 2 * L1 * L2 * (-t^4 + 6 * t^2 - 1)/(1+t^2)^2}$
$l_2 = CE = \sqrt{2 * L2^2 - 2 * L2^2 * (-t^4 + 6 * t^2 - 1)/(1+t^2)^2}$
$l_3 = AH = \sqrt{l1^2 - l2^2/4)}$
On tape :

```
L1:=143/100;L2:=154/100;
t1:=106*pi/180;t2:=109*pi/180;
t:=tan(71*pi/720);
l1:=sqrt(normal(L1^2+L2^2-2*L1*L2*(-t^4+6*t^2-1)/(1+t^2)^2));
l2:=sqrt(normal(2*L2^2-2*L2^2*(-t^4+6*t^2-1)/(1+t^2)^2));
l3:=sqrt(normal(l1^2-l2^2/4));
l4:=sqrt(normal(2*L1^2-2*L1^2*cos(t1)));
sc1:=sqrt(normal(l3^2/l1^2));
cc1:==sqrt(normal(l2^2/(4*l1^2)));
sc2:=normal(L1/l1*(4t-4t^3)/(1+t^2)^2);
cc2:==normal((L2-L1*(-t^4+6*t^2-1)/(1+t^2)^2)/l1);
cC:=coordonnees(point(0,0,0));
cE:=coordonnees(point(l2,0,0));
cA:=coordonnees(point(l2/2,l3,0));
cD:=normal([L2*(1-t^2)/(1+t^2),
    L2*2*t/(1+t^2)*(1-w^2)/(1+w^2),L2*4*t/(1+t^2)*w/(1+w^2)]);
cB:=normal(coordonnees(rotation(droite(x=0,y=0),c1,
    point(L2*cc2,-L2*sc2*(1-u^2)/(1+u^2),
          L2*sc2*2*u/(1+u^2)))));
```

```
cB:=normal(subst(subst(cB,cos(c1),cc1),sin(c1),sc1));
//cF:=normal(coordonnees(rotation(droite(x=l2,y=0),-c1,
//point(l2-L2*cc2,-L2*sc2*(1-v^2)/(1+v^2),L2*sc2*2*v/(1+v^2)))));
cF:=normal(subst([l2-cB[0],cB[1],cB[2]],u,v));
Pe:=numer((cB-cA)*(cF-cA)-L1^2*cos(t1));
Qe:=numer((cD-cE)*(cF-cE)-L2^2*(-t^4+6*t^2-1)/(1+t^2)^2);
Re:=numer(cB*cD-L2^2*(-t^4+6*t^2-1)/(1+t^2)^2);
```

mais cela donne des expressions vraiment compliquées ! ! !

Application du lemme pour le glucose

On a :
$P = \alpha_1 u^2 v^2 + \beta_1(u^2 + v^2) + \gamma_1 uv + \delta_1 = -0.918886458988 * u^2 v^2 + 1$
$1.50972814948 * u^2 + 2.96432403045 * u * v + 1.50972814948 * v^2 + 2.56689196034$
$R = \alpha u^2 w^2 + \beta u^2 + \gamma uw + +\eta w^2 \delta = (-0.696642093056) * u^2 * w^2 +$
$2.05453942839 * u^2 + 3.07941052911 * u * w + 1.94741717452 * w^2 + 3.10242421803$

— si $u = v$ alors u est égal à une des solutions $u_1, u_2, u_3 = -u_1, u_4 = -u_2$ de l'équation bicarrée :
  $P(x, x) = \alpha_1 x^4 + (2\beta_1 + \gamma_1)x^2 + \delta_1$ et w est alors solution de :
  $P(u_j, x) = 0$ pour $j = 1, 2, 3, 4$
  On tape :
  `solve(subst(P,v,u)=0,u)`
  On obtient :
  ```
  [-2.1224247841e-72-0.635547690206*i,
  -2.1224247841e-72+0.635547690206*i,
  -2.62981213076,2.62981213076]
  ```
  On tape :
  `solve(subst(R,u,2.62981213076)=0,w)`
  On obtient :
  `[-1.42146544175,4.24267305313]`
  Donc les solutions $[u, v, w]$ sont :
  ```
  [2.62981213076,2.62981213076,-1.42146544175]
  [2.62981213076,2.62981213076,4.24267305313]
  ```
  ce qui donne comme coordonnées de B, D, F :
  `u,v,w:=[2.62981213076,2.62981213076,-1.42146544175]`
  `cB` renvoie
  `[0.111689118901,1.42546684563,0.571987598416]`
  `cF` renvoie
  `[2.39578667764,1.42546684563,0.571987598416]`
  `cD` renvoie
  `[1.25373789827,-0.302153034806,-0.841691645439]`
  `u,v,w:=[2.62981213076,2.62981213076,4.24267305313]`
  `cB` renvoie
  `[0.111689118901,1.42546684563,0.571987598416]`
  `cF` renvoie
  `[2.39578667764,1.42546684563,0.571987598416]`
  `cD` renvoie
  `[1.25373789827,-0.800148906999,0.399378278169]`

`purge(u,v,w)`
Que l'on compare avec les coordonnées trouvées par la géométrie :
B de coordonnées $0.111689118901,\ 1.42546684563,\ 0.571987598417$
F de coordonnées $2.39578667764,\ 1.42546684563,\ 0.571987598417$
D de coordonnées $1.25373789827,\ -0.302153034804,\ -0.84169164544$
soit
B de coordonnées $0.111689118901,\ 1.42546684563,\ 0.571987598417$
F de coordonnées $2.39578667764,\ 1.42546684563,\ 0.571987598417$
D de coordonnées $1.25373789827,\ -0.800148906999,\ 0.399378278168$

— si $u \neq v$ on a u et v sont les solutions différentes de :
$subst(R, u, x) = (\alpha w^2 x^2 + \beta)x^2 + \gamma wx + \eta w^2 + \delta = ax^2 + bx + c$
On a :
$uv = x_1 x_2 = c/a$ et $u + v = x_1 + x_2 = -b/a$ donc
$u^2 v^2 = x_1^2 x_2^2 = c^2/a^2$ et
$u^2 + v^2 = (u + v)^2 - 2uv = x_1^2 + x_2^2 = (b^2 - 2ca)/a^2$.
On doit donc résoudre l'équation en w `subst(subst(P,u,x1),v,x2)=0` :
$\alpha_1 c^2 + \beta_1(b^2 - 2ca) + \gamma_1 ca + \delta_1 a^2 = 0$
On tape :

```
a:=alpha*w^2+beta;
b:=gamma*w;
c:=eta*w^2+delta;
```

On tape :

```
alpha1:=-0.918886458988;
beta1:=1.50972814948;
gamma1:=2.96432403045;
delta1:=2.56689196034;
alpha:=-0.696642093056;
beta:=2.05453942839;
gamma:=3.07941052911;
eta:=1.94741717452;
delta:=3.10242421803;
a:=alpha*w^2+beta;
b:=gamma*w;
c:=eta*w^2+delta;
Eq:=normal(alpha1*c^2+beta1*(b^2-2*c*a)+gamma1*c*a+
          delta1*a^2);
```

On obtient :

```
-2.16428171835*w^4-4.2362009215*w^2+1.63945920012
```

On tape :

```
solve(Eq=0,w)
```

On obtient :

```
[5.72819569555e-13+1.51272986977*i,
5.72819569555e-13-1.51272986977*i,
[-0.575349882482,0.575349882482]
```

On tape :

```
solve(subst(R,w,0.575349882482)=0,u)
```

On obtient des solutions complexes :

```
[-0.485692101424-1.34851633381*i,
```

`-0.485692101424+1.34851633381*i]`
Il n'y a donc pas d'autres solutions réelles que celles obtenues quand $u = v$.
Donc les solutions $[u, v, w]$ sont :
`[2.62981213076,2.62981213076,-1.42146544175]`
`[2.62981213076,2.62981213076,4.24267305313]`
et leurs opposées :
`[-2.62981213076,-2.62981213076,1.42146544175]`
`[-2.62981213076,-2.62981213076,-4.24267305313]`

Avec le résultant

On tape pour éliminer w des équations $Q = 0, R = 0$:
`RQ:=resultant(exact(Q),exact(R),w)`
On tape pour éliminer v des équations $P = 0, RQ = 0$:
`PRQ:=resultant(exact(P),RQ,v)`
On tape pour résoudre l'équation en u $PRQ = 0$:
`solve(PRQ=0,u)`
On trouve 2 solutions réelles pour u :
`[2.62981213079,-2.62981213079]`
On tape pour résoudre l'équation en v $\texttt{subst}(RQ, u, 2.62981213079) = 0$:
`solve(subst(RQ,u,2.62981213079)=0,v)`
On trouve pour v :
`[-1.38377770635,2.62981187867,2.62981232198,4.13642081511]`
On tape pour résoudre l'équation en w $\texttt{subst}(R, u, 2.62981213079) = 0$:
`solve(subst(R,u,2.62981213079)=0,v)`
On trouve pour w :
`[-1.42146544174,4.24267305306]`
Il reste à vérifier parmi ces solutions celles qui vérifient :
$P = 0, Q = 0, R = 0$.
`u,v,w:=2.62981213079,-1.38377770635,-1.42146544174` pas bon car
`P,Q,R` renvoie :
`-7.0571646318,14.3332486634,6.43467501504e-11`

`u,v,w:=2.62981213079,2.62981187867,-1.42146544174` bon car
`P,Q,R` renvoie :
`4.45846289665e-06,2.45798105425e-07,6.43467501504e-11`

`u,v,w:=2.62981213079,2.62981232198,-1.42146544174` bon car
`P,Q,R` renvoie :
`-3.38296679558e-06,-1.86282647974e-07,6.43467501504e-11`

`u,v,w:=2.62981213079,4.13642081511,-1.42146544174` pas bon car
`P,Q,R` renvoie :

`-37.6474413798,1.72803993337e-10,6.43467501504e-11`

`u,v,w:=2.62981213079,-1.38377770635,4.24267305306` pas bon car
`P,Q,R` renvoie :
`-32.1653712253,-83435.2258003-144.016035042*i,-9.13125437063e+14`

`u,v,w:=2.62981213079,2.62981187867,4.24267305306` bon car
`P,Q,R` renvoie :
`4.45846289665e-06,1.06109131082e-05,8.6524210019e-10`

`u,v,w:=2.62981213079,2.62981232198,4.24267305306` bon car
`P,Q,R` renvoie :
`-3.38296679558e-06,-8.04602305493e-06,-1.10418341137e-10`

`u,v,w:=2.62981213079,4.13642081511,4.24267305306` pas bon car
`P,Q,R` renvoie :
`-37.6474413798,-87.2031710848,-1.10418341137e-10`
Donc les solutions approchées sont u, v, w :
2.62981213079,2.62981187867,-1.42146544174
2.62981213079,2.62981187867,4.24267305306
On retrouve les valeurs précédentes qui étaient meilleures :
`[2.62981213076,2.62981213076,-1.42146544175]` pour lesquelles
P, Q, R=-6.62367938276e-11,-6.89368562234e-11,-6.89368562234e-11
`[2.62981213076,2.62981213076,4.24267305313]` pour lesquelles
P, Q, R=-6.62367938276e-11,1.35003119794e-11,1.35003119794e-11

15.6.5 Les commandes des figures

Pour le cyclohexane le modéle en carton
Pour avoir le modéle on tape dans un niveau de géométrie 2d :

```
a1:=71*pi/360;
l:=2*cos(a1);
A:=point(0);C:=point(l);
triangle_equilateral(A,C,E);
B:=point(l/2-i*sin(a1));
triangle_isocele(B,A,-109*pi/180);
F:=rotation(A,pi/3,point(l/2+i*sin(a1)));
triangle(A,E,F);
D:=rotation(C,-pi/3,point(l/2+i*sin(a1)));
triangle(C,E,D);
DD:=translation(C-A,B));
triangle(B,C,DD);
FF:=translation(E-C,D);
triangle(E,D,FF);
BB:=translation(A-E,F);
triangle(F,A,BB);
```

```
H:=milieu(A,C);
segment(F,D,affichage=2);H1:=milieu(F,D);
angle(A,B,C,"a1");angle(A,C,E,"a");
```

Pour le glucose le modéle en carton
Pour avoir le modéle on tape :

```
L1:=1.43;L2:=1.54;
t1:=106*pi/180.;
t2:=109*pi/180.;
c0:=71*pi/360.;
l1:=sqrt(normal(L1^2+L2^2-2*L1*L2*cos(t2)));
l2:=sqrt(normal(2*L2^2-2*L2^2*cos(t2)));
l3:=sqrt(normal(l1^2-l2^2/4));
l4:=sqrt(normal(2*L1^2-2*L1^2*cos(t1)));
A:=point(l2/2,l3);
C:=point(0);
E:=point(l2);
H:=point(l2/2);
D:=point(L2*cos(c0),-L2*sin(c0));
triangle(A,C,E);
triangle(D,C,E);
B:=inter_unique(cercle(C,L2),cercle(A,L1));
F:=inter_unique(cercle(E,L2),cercle(A,L1));
triangle(A,C,B);
triangle(A,E,F);
FF:=D+l2;
triangle(D,E,FF);
BB:=similitude(F,l4/L1,-37*pi/180,A);
triangle(A,BB,F);
DD:=similitude(B,l2/L2,-71*pi/360,C);
triangle(C,B,DD);
```

les notations
Pour avoir la figure avec les notations, on tape :

```
A:=point(l2/2,l3);
C:=point(0);
E:=point(l2);
H:=point(l2/2);
D:=point(L2*cos(c0),-L2*sin(c0));
triangle(A,C,E);
triangle(D,C,E);
angle(C,D,E,"c0");
angle(E,C,D,"c0");
angle(C,E,A,"c1");
angle(E,A,C,"c1");
B:=inter_unique(cercle(C,L2),cercle(A,L1));
F:=inter_unique(cercle(E,L2),cercle(A,L1));
```

```
triangle(A,C,B);
triangle(A,E,F);
angle(C,A,B,"c2");
angle(A,C,B,"a3");
angle(D,E,C,"t2");
angle(F,A,E,"t2");
angle(B,A,C,"t2");
angle(A,F,E,"a3");
angle(E,F,A,"c2");
legende(milieu(A,B),"L1",quadrant2);
legende(milieu(C,B),"L2",quadrant2);
legende(milieu(C,D),"L2",quadrant3);
segment(A,H);
legende(milieu(A,H),"l3",quadrant3);
legende(milieu(A,C),"l1",quadrant2);
legende(milieu(E,C),"l2",quadrant4);
BB:=point(3.3);
FF:=point(3.3+l4);
AA:=point(3.3+l4/2,L1*sin(37*pi/180),affichage=quadrant1);
triangle(AA,BB,FF);
angle(AA,BB,FF,"t1");
legende(milieu(AA,BB),"L1",quadrant2);
legende(milieu(AA,FF),"L1");
```

Chapitre 16

Les fractions continues

`Définition` Une fraction continue est une expression de la forme :
$\mathtt{a}_1 + \frac{\mathtt{b}_1}{\mathtt{a}_2 + \frac{\mathtt{b}_2}{\mathtt{a}_3 + \frac{\mathtt{b}_3}{\mathtt{a}_4 + ...}}}$
avec pour $k > 1$ $a_k > 0$ et pour $k \geq 1$ $b_k > 0$ Une fraction continue simple est une fraction continue où les $b_k = 1$, c'est donc une expression de la forme :
$\mathtt{a}_1 + \frac{1}{\mathtt{a}_2 + \frac{1}{\mathtt{a}_3 + \frac{1}{\mathtt{a}_4 + ...}}}$
Ici on ne s'intéressera qu'aux fractions continues simples car sinon l'écriture n'est pas unique, par exemple : $\sqrt{13} = 3 + \frac{4}{6 + \frac{4}{6 + \frac{4}{6 + ...}}} =$
$2 + \frac{9}{4 + \frac{9}{4 + \frac{9}{4 + ...}}} =$
$3 + \frac{1}{1 + \frac{1}{1 + \frac{1}{1 + + \frac{1}{1 + + \frac{1}{6 + ...}}}}}$
`Notation` Si $\mathtt{n} = \mathtt{a}_1 + \frac{1}{\mathtt{a}_2 + \frac{1}{\mathtt{a}_3 + \frac{1}{\mathtt{a}_4 + ...}}}$
on notera :
$n = ([a_1, a_2, ...a_n, ..], [])$ si le développement n'est pas périodique et
$n = ([a_1, a_2, ...a_s], [b_1, b_2, ...b_t])$ si le développement est périodique de période $[c_1, c_2, ...c_t]$ ie lorsque $n = [a_1, a_2, ...a_s, c_1, c_2, ...c_t, c_1, c_2, ...c_t, c_1....]$
Par exemple :
$\sqrt{13} = ([3, 1, 1, 1, 1, 6], [1, 1, 1, 1, 6])$

`Propriétés`
Un nombre rationnel a un développement en fraction continue fini.
Les réels qui ont un développement en fraction continue périodique sont solution d'une équation du second degré à coefficients dans $\mathbb{N}$.
`Le programme :`

```
f2dfc(x,n):={
local r,q,lq,lr,p,j;
q:=floor(x);
r:=normal(x-q);
lq:=[];
lr:=[];
for (j:=1;j<=n;j:=j+1) {
lq:=concat(lq,q);
if (x==q){return (lq,[]);}
```

```
p:=member(r,lr);
if (p) {return (lq,mid(lq,p))};
lr:=concat(lr,r);
x:=normal(1/r);
q:=floor(x);
r:=normal(x-q);
}
return (concat(lq,x),[]);
};

dfc2f1(d,t):={
local s,st,x,l,xt,k;
s:=size(d);
x:=d[s-1];
for (k:=s-2;k>=0;k:=k-1) {x:=normal(d[k]+1/x);}
if (t==[]) {return normal(x);}
st:=size(t);
purge(y);
xt:=t[st-1]+y;
for (k:=st-2;k>=0;k:=k-1) {xt:=normal(t[k]+1/xt);}
l:=solve(y=1/xt,y);
if (l[0]>0){y:=normal(l[0]);}else{y:=normal(l[1]);};
x:=d[s-1]+y;
for (k:=s-2;k>=0;k:=k-1) {x:=normal(d[k]+1/x);}
return(normal(x));
};
dfc2f2(d,t):={
local s,st,x,xt;
s:=size(d);
x:=d[s-1];
for (k:=s-2;s>=0;s:=s-1) {x:=d[k]+1/x;}
if (t==[]) {return x;}
st:=size(t);
xt:=t[st-1];
for (k:=st-2;st>=0;st:=st-1) {xt:=st[k]+1/xt;}
return(x+1/2*(sqrt(xt^2+4)-xt));
};
```

Chapitre 17

Pour s'amuser avec des dessins récursifs

17.1 Les sapins

```
sapin(x,y):={
 DispG();
 if (abs(x-y)<0.5) {segment(x,y); return 0;}
 sapin(x,x+(y-x)*0.5*exp(i));
 sapin(x,x+(y-x)*0.5*exp(-i));
 segment(x,(3*x+y)/4);
 sapin((3*x+y)/4,y);
}
```

je voulais utiliser similitude mais je n'arrive pas a me debarasser des petites croix qui marque le point..meme en faisant nodisp(similitude(....))

```
sapinp(x,y):={
 DispG();
 if (abs(x-y)<0.2) {segment(x,y); return 0;}
 sapin(x,affixe(similitude(x,0.5,1.0,y)));
 sapin(x,affixe(similitude(x,0.5,-1.0,y)));
 segment(x,(3*x+y)/4);
 sapin((3*x+y)/4,y);
}:;
```

Sans utiliser l'écran `DispG` on met toutes les instructions dans une séquence.
On tape :

```
sapin(z0,z1):={
  local L,v;
    L:=NULL;
    v:=z1-z0;
    si abs(v)<0.5 alors L:=L, segment(z0,z1);retourne L; fsi;
    L:=L,sapin(z0+v/4,z1);
    L:=L,segment(z0,z0+v*0.25);
    L:=L,sapin(z0,z0+v*exp(i*pi/6.)*0.5);
```

```
    L:=L,sapin(z0,z0+v*exp(-i*pi/6.)*0.5);
    }:;
```

ou

```
sapin1(z0,z1,t):={
  local L,v;
    L:=NULL;
    v:=z1-z0;
    si abs(v)<0.5 alors L:=L, segment(z0,z1);retourne L; fsi;
    L:=L,sapin1(z0+v/4,z1,t);
    L:=L,segment(z0,z0+v*0.25);
    L:=L,sapin1(z0,z0+v*exp(i*t)*0.5,t);
    L:=L,sapin1(z0,z0+v*exp(-i*t)*0.5,t);
    }:;
```

17.2 Les fleurs

```
fleur(x,y):={
 DispG();
 if (abs(x-y)<0.5) {segment(x,y);cercle(y,(y-x)*0.3); return 0;}
 segment(x,y);cercle(y,(y-x)*0.3);cercle(y,(y-x)*0.2);
 fleur(x,x+(y-x)*0.5*exp(i*0.5));
 fleur(x,x+(y-x)*0.5*exp(-i*0.5));
}
```

On tape :
`fleur(0,5*i)`
On obtient le dessin dans `DispG`.

17.3 L'anémone

```
anemone(a,n,c):={
local L;
L:=NULL;
L:=L,cercle(a+i*n,n);
L:=L,cercle(a-i*n,n);
L:=L,cercle(a+n,n);
L:=L,cercle(a-n,n);
retourne affichage(L,c+rempli),cercle(a,n/2,affichage=rempli+0);
}:;
```

On tape :
`anemone(0,5,1)`
On obtient

17.4 La tulipe

```
tulipe(a,n,c):={
local L;
L:=NULL;
L:=L,cercle(a+i*n,n,3.15,6.4);
L:=L,cercle(a+i*n-n,n,-0.1,1.58);
L:=L,cercle(a+i*n+n,n,1.58,3.15);
retourne affichage(L,c+rempli);
}:;
```

On tape :
`tulipe(0,5,3)`
On obtient

17.5 Les bouquets de fleurs

Ici on veut avoir le dessin comme réponse, on met donc toute la suite des instructions graphiques dans une liste `L` qui sera retournée comme réponse.
On tape pour avoir un bouquet d'anémones :

```
fleurs1(x,y):={
 local L,c;
 L:=NULL;
 c:=alea(10)+85;
 if (abs(x-y)<0.5) {return L,arc(x,y,-5*pi/12,
     affichage=epaisseur_ligne_3),anemone(y,(y-x)*0.3,c); }
 L:=arc(x,y,-5*pi/12,affichage=epaisseur_ligne_3),anemone(y,(y-x)*0.2,c);
 L:=L,fleurs1(x,x+(y-x)*0.5*exp(i*0.5));
 L:=L,fleurs1(x,x+(y-x)*0.5*exp(-i*0.5));
retourne L;
}:;
```

On tape :
`fleurs1(0,5*i)`
On tape pour une carte de "bonne annee" :
```
fleurs1(0,5*i),fleurs1(0,4+4*i),legende(10+5i,"BONNE "),
legende(10+3i,"ANNEE"),legende(10.2+i,"2011")
```
On tape pour avoir un bouquet de tulipes :

```
fleurs3(x,y):={
 local L,c;
 L:=NULL;
 c:=alea(10)+85;
 if (abs(x-y)<0.5) {return L,arc(x,y,-5*pi/12,
        affichage=epaisseur_ligne_3),
        tulipe(y,(-i)*(y-x)*(0.5)*exp(-i*5*pi/24),c);}
 L:=arc(x,y,-5*pi/12,affichage=epaisseur_ligne_3),
        tulipe(y,-i*(y-x)*(0.4)*exp(-i*5*pi/24),c);
```

```
 L:=L,fleurs3(x,x+(y-x)*0.6*exp(i*0.5));
 L:=L,fleurs3(x,x+(y-x)*0.6*exp(-i*0.5));
 retourne L;
}:;
```

On tape :
`fleurs3(0,5*i)`
On tape pour une carte de "bonne annee" :
```
fleurs3(0,5*i),fleurs3(0,3+4*i),fleurs3(0,-3+4*i),
legende(10+7i,"BONNE "),legende(10+5i,"ANNEE"),
legende(10.2+3i,"2011")
```
On tape :
```
fleurs3(0,5*i),fleurs3(0,3+4*i),fleurs1(14,14+6*i),
fleurs1(14,19+6*i),legende(8.2+7i,"BONNE "),
legende(8.2+5i,"ANNEE"),legende(8.4+3i,"2011")
```

17.6 La fougère

Ce n'est pas un dessin récursif, mais c'est une une suite d'itérées. Pour obtenir cette fougère, on va itérer un système de fonctions avec poids aléatoires. On considère : `A:=[[0,0,0.5],[0,0.16,0],[0,0,1]]` de poids 0.01,
`B:=[[0.2,-0.26,0.4],[0.23,0.22,0.005],[0,0,1]]` de poids 0.07,
`C:=[[-0.15,0.28,0.57],[0.26,0.24,-0.12],[0,0,1]]` de poids 0.07,
`D:=[[0.85,0.04,0.08],[-0.04,0.85,0.18],[0,0,1]]` de poids 0.85,
On définit `F:=A,B,C,D` et on itère en partant du vecteur `w0:=[0,5,1]`. Pour définir `w1`, on applique à `w0`, soit `A`, soit `B`, soit `C` ou `D` de façon aléatoire selon leur poids et on définit `w2` etc...On dessine les points définit par les 2 premières coordonnées des `w`, mais, pour la beauté du dessin, on ne dessine pas les `n1` premiers points. Donc dans fougere on a `n1` points invisibles et `n2` itérations .
On tape :

```
choisirk():={
local r,p,k;
p:=[0.01,0.07,0.07,0.85];
r:=rand(0,1);
k:=0;
tantque r>p[k] faire
  r:=r-p[k];
  k:=k+1;
ftantque;
return k;
}:;
fougere1(n1,n2):={
local A,B,C,D,F,j,k,w,P;
A:=[[0,0,0.5],[0,0.16,0],[0,0,1]];
B:=[[0.2,-0.26,0.4],[0.23,0.22,0.005],[0,0,1]];
```

```
C:=[[-0.15,0.28,0.57],[0.26,0.24,-0.12],[0,0,1]];
D:=[[0.85,0.04,0.08],[-0.04,0.85,0.18],[0,0,1]];
F:=A,B,C,D;
w:=[0,5,1];
P:=NULL;
pour j de 1 jusque n1 faire
  w:=F[choisirk()]*w;
fpour;
pour j de n1+1 jusque n2 faire
  w:=F[choisirk()]*w;
  P:=P,point(w[0],w[1]);
fpour;
return P;
}:;
```

On tape : `affichage(2+point_point);fougere1(10,10000)`
On obtient :

17.7 Une autre fougère

Une application affine f de $\mathbb{R}^2$ dans $\mathbb{R}^2$ est définie par une matrice carrée A de dimension 2 et par un vecteur colonne U de de $\mathbb{R}^2$, on a :
$f(X) = A * X + U$ pour tout vecteur colonne X de $\mathbb{R}^2$.
Pour faire cette nouvelle fougère, on fera comme précédemment les itérées d'un point (ici l'origine) en utilisant l'une des 4 applications affines de $\mathbb{R}^2$ dans $\mathbb{R}^2$ avec une probabilité donnée :

- f_1 sera utilisée avec la probabilité $p1 = 0.01$ et est définie par :
  `A1:=[[0,0],[0,0.16]]` et
  `U1:=[0,0]`
- f_2 sera utilisée avec la probabilité $p2 = 0.07$ et est définie par :
  `A2:=[[-0.15,0.28],[0.26,0.24]]` et
  `U2:=[0,0.44]`

— f_3 sera utilisée avec la probabilité $p3 = 0.07$ et est définie par :
`A3:=[[0.2,-0.26],[0.23,0.22]]` et
`U3:=[0,1.6]`
— f_4 sera utilisée avec la probabilité $p4 = 0.85$ et est définie par :
`A4:=[[0.85,0.04],[-0.04,0.85]]` et
`U4:=[0,1.6]`

On tape :

```
choisirk():={
local r,p,k;
p:=[0.01,0.07,0.07,0.85];
r:=rand(0,1);
k:=0;
tantque r>p[k] faire
  r:=r-p[k];
  k:=k+1;
ftantque;
return k;
}:;
fougere2(n):={
local A,A1,A2,A3,A4,U,U1,U2,U3,U4,j,k,P,w;
A1:=[[0,0],[0,0.16]];
U1:=[0,0];
A2:=[[-0.15,0.28],[0.26,0.24]];
U2:=[0,0.44];
A3:=[[0.2,-0.26],[0.23,0.22]];
U3:=[0,1.6];
A4:=[[0.85,0.04],[-0.04,0.85]];
U4:=[0,1.6];
A:=A1,A2,A3,A4;
U:=U1,U2,U3,U4;
w:=[0,0];
P:=NULL;
pour j de 1 jusque n faire
  k:=choisirk();
  w:=A[k]*w+U[k];
  P:=P,point(w[0],w[1]);
fpour;
return P;
}:;
```

On tape : `affichage(2+point_point);fougere2(3000)`
On obtient :

Remarque Ces 2 fougères sont sensiblement les mêmes : la première est décrite en coordonnèes homogènes. Voici le programme `fougere3` de la fougère2 en coordonnèes homogènes. Ce programme est un peu plus lent

```
choisirk():={
local r,p,k;
p:=[0.01,0.07,0.07,0.85];
r:=rand(0,1);
k:=0;
tantque r>p[k] faire
  r:=r-p[k];
  k:=k+1;
ftantque;
return k;
}:;
fougere3(n):={
local A,A1,A2,A3,A4,j,k,P,w;
A1:=[[0,0,0],[0,0.16,0],[0,0,1]];
A2:=[[-0.15,0.28,0],[0.26,0.24,0.44],[0,0,1]];
A3:=[[0.2,-0.26,0],[0.23,0.22,1.6],[0,0,1]];
A4:=[[0.85,0.04,0],[-0.04,0.85,1.6],[0,0,1]];
A:=A1,A2,A3,A4;
w:=[0,0,1];
P:=NULL;
pour j de 1 jusque n faire
  k:=choisirk();
  w:=A[k]*w;
  P:=P,point(w[0],w[1]);
fpour;
return P;
}
:;
```

17.8 Le choux-fleur

17.8.1 L'énoncé

Soit la suite de fonctions f_n de [0 ;1] dans $\mathbb{R}$ définie par :
pour $n = 1$:

$$f_1(x) = \begin{cases} x & \text{si } x \leq \frac{1}{2} \\ 1 - x & \text{si } x > \frac{1}{2} \end{cases}$$

et pour $n \geq 2$:

$$f_n(x) = \begin{cases} \frac{1}{2} f_{n-1}(2x) & \text{si } x \leq \frac{1}{2} \\ f_n(x - \frac{1}{2}) & \text{si } x > \frac{1}{2} \end{cases}$$

La fonction f_n est périodique de période $\frac{1}{2^{n-1}}$ et tracer son graphe sur $[0; \frac{1}{2^{n-1}}]$.

1. Définir dans `Xcas` une fonction calculant $f_n(x)$.
2. Définir dans `Xcas` une fonction calculant la somme :
   $S_n(x) = f_1(x) + f_2(x) + ...f_n(x)$
3. Tracer le graphe de S_{10}
4. On définit le dessin suivant :
   Soient deux points $A(x_a, y_a)$ et $B(x_b, y_b)$ à partir du vecteur AB, on construit les vecteurs AC et CB où C a comme coordonnées $x_c = (x_a + x_b)/2$ et $y_c = (y_a + y_b)/2 + (x_b - x_a)/2$. On recommence la même construction à partir de chacun des 2 vecteurs obtenus etc....
   Écrire un programme qui dessine les segments obtenus au bout de n fois (mais pas les dessins intermédiaires). On remarquera que l'on obtient ainsi le graphe de S_n.

17.8.2 La correction

1. On définit la fonction `f(n,x)` :

```
f(n,x):={
si x<=1/2 alors
si n==1 alors return x;sinon return f(n-1,2x)/2; fsi;
fsi
si n==1 alors return 1-x;sinon return f(n,x-1/2); fsi;
}:;
```

2. On tape : `plotfunc(f(4,x),x=0..1,affichage=1)`
   On obtient :



3. On définit la fonction `S(n,x)` :

```
S(n,x):={
local j,s;
pour j de 1 jusque n faire
s:=s+f(j,x);
fpour;
return s;
}:;
```

4. On définit la fonction récursive `chouxfleur(A,B,n)` :

```
chouxfleur(A,B,n):={
  local xa,ya,xb,yb,xc,yc,C;
  si n==0 alors return segment(A,B);fsi;
  xa,ya:=coordonnees(A);
  xb,yb:=coordonnees(B);
  xc:=(xa+xb)/2;
  yc:=(ya+yb)/2+(xb-xa)/2;
  C:=point(xc,yc);
  return chouxfleur(A,C,n-1),chouxfleur(C,B,n-1);
}:;
```

On tape : `chouxfleur(point(0),point(1),10)`
On obtient :



17.9 Les arbres

17.9.1 Un arbre

```
arbre(x,y):={
 DispG();
 if (abs(x-y)<0.2) {segment(x,y); return 0;}
 segment(x,(x+y)/2);
 arbre((x+y)/2,(x+y)/2+(y-x)*0.5*exp(i*0.5));
 arbre((x+y)/2,(x+y)/2+(y-x)*0.5*exp(-i*0.5));
}
```

17.9.2 Un arbre moins déplumé

```
arbre2(x,y):={
```

```
 DispG();
 if (abs(x-y)<0.2) {segment(x,y); return 0;}
 segment(x,(x+y)/2);
 arbre2((x+y)/2,(x+y)/2+(y-x)*0.5*exp(i*0.5));
 arbre2((x+y)/2,(x+y)/2+(y-x)*0.5*exp(-i*0.5));
 arbre2((x+y)/2,(x+y)/2+(y-x)*0.5*exp(i));
 arbre2((x+y)/2,(x+y)/2+(y-x)*0.5*exp(-i));
}
```

17.9.3 Un arbre epineux

```
arbre3(x,y):={
 DispG();
 if (abs(x-y)<0.2) {segment(x,y); return 0;}
 segment(x,(x+y)*0.5);
 arbre3((3*x+y)/4,(3*x+y)/4+(y-x)*0.25*exp(i*0.5));
 arbre3((3*x+y)/4,(3*x+y)/4+(y-x)*0.25*exp(-i*0.5));
 arbre3((x+y)/2,(x+y)/2+(y-x)*0.5*exp(i));
 arbre3((x+y)/2,(x+y)/2+(y-x)*0.5*exp(-i));
}
```

17.9.4 Le bouquet final

```
bouquet(x,y):={
 DispG();
 if (abs(x-y)<0.2) {segment(x,y); return 0;}
 segment(x,(x+y)*0.5);
 bouquet((3*x+y)/4,(3*x+y)/4+(y-x)*0.25*exp(i*0.5));
 bouquet((3*x+y)/4,(3*x+y)/4+(y-x)*0.25*exp(-i*0.5));
 bouquet((x+y)/2,(x+y)/2+(y-x)*0.5*exp(i));
 bouquet((x+y)/2,(x+y)/2+(y-x)*0.5*exp(-i));
 bouquet((x+y)/2,(x+y)/2+(y-x)*0.5);
```

17.10 Un exercice tiré des olympiades

17.10.1 L'énoncé

On modélise la croissance d'un arbre de la manière suivante : au départ il est composé d'un tronc vertical AB de longueur l, au bout d'un an ce tronc donne naissance a 2 branches BC et BD vérifiant : $BC = 0.5 * AB$, $BD = 0.75 * AB$, et
$(\overrightarrow{BC}, \overrightarrow{BA}) = (\overrightarrow{BA}, \overrightarrow{BD}) = 5\pi/6$.
Chaque année, chaque branche donne naissance a 2 branches selon le même processus. Faire le dessin de l'arbre au bout de 3 ans
Si on ne tient pas compte de la durée de vie de l'arbre, quelle est la limite de sa taille ?

Avec `Xcas` on pourra exécuter `arbre.xws` pour avoir la correction.

17.10.2 Le dessin

Pour faire le dessin on écrit une procédure récursive `arbre` qui renvoie 0 quand elle se termine et qui a comme paramètres : `A` le point de plantation, `l` la longueur du tronc `AB`, `t` l'angle que fait `AB` avec le sol, `n` le nombre d'années. Ainsi le point `B` a comme affixe `A+l*exp(i*t)`.
On verra le dessin dans l'écran `DispG` (`session->montrer->DispG`).

```
arbre(A,l,t,n):={
  local B;
  B:=A+l*exp(i*t);
  segment(A,B);
  if (n>0){
    arbre(B,l*0.5,t+pi/6,n-1);
    arbre(B,l*0.75,t-pi/6,n-1);
  }
  return 0;
}:;
```

17.10.3 La taille

Pour connaitre la taille de l'arbre au bout de `n` années, on écrit une procédure récursive `harbre` qui a les mêmes paramètres que la procédure récursive `arbre` et qui renvoie l'ordonnée exacte du point le plus haut de l'arbre.

```
harbre(A,l,t,n):={
  local B,res;
  B:=A+l*exp(i*t);
  res:=max(ordonnee(A),ordonnee(B));
  if (n>0){
    res:=normal(max(res,harbre(B,l/2,t+pi/6,n-1),
    harbre(B,l*3/4,t-pi/6,n-1)));
  }
  return res;
}:;
```

ou plutôt pour éviter des calculs trop longs on écrit `harbra` qui renvoie la valeur approchée de l'ordonnée du point de l'arbre le plus haut. On peut aussi utiliser `harbre` en mettant comme valeur de `l` une valeur décimale : par exemple `1.0`.

```
harbra(A,l,t,n):={
local B,res;
B:=evalf(A+l*exp(i*t));
res:=max(ordonnee(A),ordonnee(B));
if (n>0){
res:=max(res,harbre(B,l*0.5,evalf(t+pi/6),n-1),
harbre(B,l*0.75,evalf(t-pi/6),n-1));
}
return res;
}:;
```

Pour avoir la hauteur de l'arbre, il faut soit planter l'arbre en un point `A` d'ordonée `0`, soit utiliser la procédure `hauteur_arbre` ci-dessous qui suppose que le tronc de l'arbre est vertical :
`hauteur_arbre(A,l,n):=harbre(A,l,pi/2,n)-ordonnee(A):;` On tape : `harbre(0,1,pi/2,10)` On obtient au bout de 30s :
`(19515*sqrt(3)+52283)/32768`
On tape : `harbre(0,1.0,pi/2,1)` On obtient au bout de 5s :
`2.62707432586`
On tape : `hauteur_arbre(i,1.0,pi/2,1)` On obtient au bout de 5s :
`2.62707432586`

17.10.4 La limite de la taille

Seulement, cela ne nous donne qu'une valeur approchée de la limite de cette hauteur.
Pour connaitre la taille maximum h de l'arbre, il faut remarquer que :
- si l'arbre poussait verticalement sa hauteur serait une série géométrique de raison 3/4 de somme 4
- la hauteur d'un arbre est proportionnelle à la longueur du tronc initial,
- qu'au bout de 2 ans la branche gauche de la branche droite est verticale et est de longueur $l * 3/4 * /2 = 3l/8$ c'est à dire que cette branche est de hauteur les $3h/8$.

On calcule à la main la hauteur de l'arbre au bout d'1 an : $l + (\sqrt{3} * l/2) * (3/4) = l(1 + 3\sqrt{3}/8)$ ou on tape :
`harbre(0,1,pi/2,1);harbre(0,1.0,pi/2,1);`
On obtient :
`(3*sqrt(3)+8)/8,1.64951905284`
On a donc l'équation :
$h = 1 + (\sqrt{3}/2) * (3/4) + 3 * h/8$
On tape :
`solve(h=3/8*h+(1+3*sqrt(3)/8) ,h)`
On obtient :
`[1/5*(3*sqrt(3)+8)]`
On tape :
`evalf(1/5*(3*sqrt(3)+8))`
On obtient :
`2.63923048454`
La taille maximum de l'arbre de tronc l est :
$l * (3 * sqrt(3) + 8)/5 \simeq 2.63923048454 * l$

Chapitre 18

Les carrés magiques

18.1 Les matrices magiques d'odre 3

18.1.1 Résultat préliminaire

Soit A la matrice :
$A = [[a_{00}, a_{01}, a_{02}], [a_{10}, a_{11}, a_{12}], [a_{20}, a_{21}, a_{22}]]$ Montrer que :
$B = 1/2 * (A + \text{tran}(A))$ est une matrice symértrique et
$C = 1/2 * (A - \text{tran}(A))$ est une matrice antisymértrique.
En déduire que toute matrice se décompose de façon unique en la somme d'une matrice symétrique et d'une matrice antisymétrique.
On tape :

```
A:=[[a00,a01,a02],[a10,a11,a12],[a20,a21,a22]]
B:=1/2*(A+tran(A));C:=1/2*(A-tran(A))
normal(B-tran(B)),normal(C+tran(C))
```

On obtient :

```
[[0,0,0],[0,0,0],[0,0,0]],[[0,0,0],[0,0,0],[0,0,0]]
```

Si $A = M + N$ avec tran(M)=M et tran(N)=$-N$ alors tran(A)=$M - N$ donc
$M = 1/2 * (A + \text{tran}(A)) = B$ et $N = 1/2 * (A - \text{tran}(A)) = C$ d'où l'unicité.

18.1.2 Les matrices magiques d'ordre 3

Une matrice $A = (a_{j,k})$ d'ordre 3 est une matrice magique de somme s lorsque les 8 sommes :
$\sum_{j=0}^{2} a_{j,k} = s$ pour $k = 0, 1, 2$ (somme de chaque colonne)
$\sum_{k=0}^{2} a_{j,k} = s$ pour $j = 0, 1, 2$ (somme de chaque ligne)
$\sum_{j=0}^{2} a_{j,j} = a_{1,3} + a_{2,2} + a_{3,1} = s$ (somme de chaque diagonale).
Déterminer les matrices magiques d'ordre 3 qui sont antisymétriques.
Déterminer les matrices magiques d'ordre 3 qui sont symétriques de somme $s = 0$.
Déterminer une matrice magique d'ordre 3, symétrique, de somme s et la plus simple possible.
Montrer que la différence de 2 matrices symétriques magiques de somme s est une matrices symétriques magiques de somme $s = 0$.
En déduire toutes les matrices magiques d'ordre 3, symétriques de somme s.
Trouver toutes les matrices magiques d'ordre 3 de somme $s = 3$.
Trouver toutes les matrices magiques d'ordre 3 de somme $s = 9$.

Les matrices magiques d'ordre 3 qui sont antisymétriques sont de somme $s = 0$ car la diagonale principale ne contient que des 0.
Soit A une matrice antisymétrique d'ordre 3 et magique de somme $s = 0$.
On tape :
`A:=[[0,a,b],[-a,0,c],[-b,-c,0]]`
`linsolve([a+b=0,-a+c=0,b+c=0],[a,b,c])`
On obtient :
`[c,-c,c]`
Donc les matrices magiques A d'ordre 3 qui sont antisymétriques sont :

$$A = \begin{pmatrix} 0 & c & -c \\ -c & 0 & c \\ c & -c & 0 \end{pmatrix} = c \begin{pmatrix} 0 & 1 & -1 \\ -1 & 0 & 1 \\ 1 & -1 & 0 \end{pmatrix}$$

On tape :
`A:=c*[[0,1,-1],[-1,0,1],[1,-1,0]]`
Soit S une matrice symétrique d'ordre 3 et magique de somme $s = 0$.
On tape :
`S:=[[a,b,c],[b,d,e],[c,e,f]]`
`linsolve([a+b+c,a+d+f,b+d+e,c+e+f,2c+d],[a,b,c,d,e,f])`
On obtient :
`[-f,f,0,0,-f,f]`
Donc les matrices magiques S d'ordre 3 qui sont symétriques de somme $s = 0$ sont :

$$S = \begin{pmatrix} -f & f & 0 \\ f & 0 & -f \\ 0 & -f & f \end{pmatrix} = f \begin{pmatrix} -1 & 1 & 0 \\ 1 & 0 & -1 \\ 0 & -1 & 1 \end{pmatrix}$$

Soit S_0 une matrice symétrique d'ordre 3 et magique de somme s la plus simple possible.
On tape :
`S0:=s/3*[[1,1,1],[1,1,1],[1,1,1]]`
Donc les matrices magiques S d'ordre 3 qui sont symétriques de somme s sont $B := S + S_0$:
On tape :
`f*[[-1,1,0],[1,0,-1],[0,-1,1]]+s/3*[[1,1,1],[1,1,1],[1,1,1]]`

Donc les matrices magiques M d'ordre 3 qui sont de somme $s = 3$ (resp $s = 12$) sont $M := A + S + S_0$ et elles dépendent de 2 paramètres c et f.
On tape :
```
M:=c*[[0,1,-1],[-1,0,1],[1,-1,0]]+
f*[[-1,1,0],[1,0,-1],[0,-1,1]]+[[1,1,1],[1,1,1],[1,1,1]]
```
On obtient :
`[[-f+1,c+f+1,-c+1],[-c+f+1,1,c-f+1],[c+1,-c-f+1,f+1]]`
Par exemple $f = 1, c = 0$, on obtient `[[0,2,1],[2,1,0],[1,0,2]]`.
On tape :
```
M:=c*[[0,1,-1],[-1,0,1],[1,-1,0]]+
f*[[-1,1,0],[1,0,-1],[0,-1,1]]+4*[[1,1,1],[1,1,1],[1,1,1]]
```
On obtient :
`[[-f+4,c+f+4,-c+4],[-c+f+4,4,c-f+4],[c+4,-c-f+4,f+4]]`
Par exemple $f = 1, c = 3$, on obtient un carré magique eulérien (les entiers 0..8 apparaissent dans les 9 cases du carré) : `[[3,8,1],[2,4,6],[7,0,5]]`.

18.2 Les carrés magiques eulériens et latins pandiagonaux

Ce qui suit a èté inspiré par une épreuve à l'oral de l'agrégation externe de Mathématiques session 2005 dont voila le lien :
http ://agreg.dnsalias.org/Textes/561.pdf
et par le livre Problèmes plaisants et délectables par Claude-Gaspar Bachet sieur de Mérignac.

18.2.1 Définitions

carré magique d'ordre n veut dire que l'on a une matrice carrèe de dimension n, à coefficients dans $\mathbb{N}$ et telle que les sommes des éléments de chaque ligne, de chaque colonne et de chacune des 2 diagonales sont égales :
si $A = a_{j,k}$ pour $j = 0..n-1, k = 0..n-1$ on a :
$\sum_{k=0}^{n-1} a_{j,k} = s$ pour $j = 0..n-1$
$\sum_{j=0}^{n-1} a_{j,k} = s$ pour $k = 0..n-1$
$\sum_{j=0}^{n-1} a_{j,j} = s$
$\sum_{j=0}^{n-1} a_{j,n-j-1} = s$
carré latin d'ordre n veut dire que les entiers $0..n-1$ apparaissent dans chaque ligne, dans chaque colonne et dans les 2 diagonales du carré. La somme s est alors égale à $n(n-1)/2$
carré latin pandiagonal d'ordre n veut dire que l'on a un carré magique latin qui possède en plus la propriété : les entiers $0..n-1$ apparaissent aussi dans les n diagonales brisées du carré (lorsqu'on considère le carré comme un tore). La somme s est égale à $n(n-1)/2$
carré eulérien d'ordre n veut dire que chacun des entiers $0..n^2-1$ apparaissent dans les n^2 cases du carré. La somme s est alors égale à $n(n^2-1)/2$
carré eulérien pandiagonal d'ordre n veut dire que l'on a un carré magique eulérien d'odre n qui possède en plus la propriété : la somme de chacune des $2n-2$ diagonales brisées (lorsqu'on considère le carré comme un tore) vaut aussi $n(n^2-1)/2$.

18.2.2 Les carrés magiques latins pandiagonaux

Écrire un programme `Xcas` qui teste si une matrice est un carré magique latin pandiagonal.
On tape dans un éditeur de programme :

```
estlatinp(A):={
local s,n,j,k,L;
n:=size(A);
pour j de 0 jusque n-1 faire
L:=A[j];
si is_permu(L)==0 alors retourne faux; fsi;
fpour;
pour j de 0 jusque n-1 faire
L:=col(A,j);
si is_permu(L)==0 alors retourne faux; fsi;
fpour;
```

```
pour j de 0 jusque n-1 faire
L:=A[k,irem(j+k,n)]$(k=0..n-1);
si is_permu(L)==0 alors retourne faux; fsi;
fpour;
pour j de 0 jusque n-1 faire
L:=A[k,irem(j-k,n)]$(k=0..n-1);
si is_permu(L)==0 alors retourne faux; fsi;
fpour;
retourne vrai;
}
:;
```

18.2.3 Les carrés magiques eulériens pandiagonaux

Écrire un programme Xcas qui teste si une matrice est un carré magique eulérien pandiagonal.
On tape dans un éditeur de programme :

```
esteulerp(A):={
local n,s,j,k,rep,L,C,D,M;
L:=mat2list(A);
si is_permu(L)==0 alors retourne faux fsi;
rep:=true;
n:=size(A);
s:=n*(n^2-1)/2;
L:=unapply(sum(A[j,k],k=0..n-1),j);
C:=unapply(sum(A[j,k],j=0..n-1),k);
D:=unapply(sum(A[j,irem(j+k,n)],j=0..n-1),k);
M:=unapply(sum(A[j,irem(k-j,n)],j=0..n-1),k);
pour j de 0 jusque n faire
  si L(j)!=s ou C(j)!=s ou D(j)!=s ou M(j)!=s alors
    rep:=faux; break;
  fsi;
fpour;
retourne rep;
}
:;
```

On remarque que si les n lignes et $n-1$ colonnes (resp n lignes et $n-1$ diagomales montantes, n lignes et $n-1$ diagomales descendantes) sont de même somme s alors les n colonnes (resp les n diagomales montantes, les n diagomales descendantes) sont de même somme s. On tape plus simplement pour savoir si A est un carré eulérien pandiagonal :

```
estpandiage(A):={
local j,k,n,s,L;
L:=mat2list(A);
si is_permu(L)==0 alors retourne faux fsi;
```

```
n:=size(A);
s:=n*(n^2-1)/2;
si [sum(A[j,k],j,0,n-1)$(k=0..n-1)]!=[s $(k=0..n-1)] alors
retourne faux;
fsi;
si [sum(A[k,j],j,0,n-1)$(k=0..n-2)]!=[s $(k=0..n-2)] alors
retourne faux;
fsi;
si [sum(A[j,irem(j+k,n)],j,0,n-1)$(k=0..n-2)]!=[s $(k=0..n-2)] alors
retourne faux;
fsi;
si [sum(A[j,irem(-j+k,n)],j,0,n-1)$(k=0..n-2)]!=[s $(k=0..n-2)] alors
retourne faux;
fsi;
retourne vrai;
}
:;
```

18.2.4 La règle de Manuel Moscopule (14ième siécle) ou règle du cavalier

Cette règle permet d'écrire des carrés magiques eulériens pandiagonaux d'ordre n lorsque n est impair et non divisible par 3.
On considère le carré comme un tore et on applique la règle :

- Placez 0 dans la case de votre choix (case l, c du programme),
- puis, avancez horizontalement de 2 cases et descendez verticalement d'une case et placez le 1 (mouvement du cavalier aux echecs),
- faites de même pour les nombres suivants jusqu'au placement de $n-1$,
- reculez horizontalement de 1 case et placez alors le nombre n,
- refaites le même processus jusque $2n-1$, puis reculez horizontalement de 1 case et placez alors le nombre $2n$ etc...

Écrire un programme `Xcas` qui renvoie un carré magique eulérien construit selon cette règle.
On tape dans un éditeur de programme :

```
cavalier(n,l,c):={
local j,k,A;
si irem(n,2)==0 ou irem(n,3)==0
  alors retourne "n !=2k et n !=3k";
fsi
l:=irem(l,n);c:=irem(c,n);
A:=idn(n);
pour j de 0 jusque n-1 faire
  pour k de 0 jusque n-2 faire
    A[l,c]:=k+n*j;
    l:=irem(l+1,n);
    c:=irem(c+2,n);
  fpour;
A[l,c]:=k+n*j;
```

```
c:=irem(c-1,n);
fpour;
return A;
}
:;
```

On tape : `C:=cavalier(5,0,0)`
On obtient :
`[[0,24,18,12,6],[13,7,1,20,19],[21,15,14,8,2],`
`[9,3,22,16,10],[17,11,5,4,23]]`
c'est à dire :

0	24	18	12	6
13	7	1	20	19
21	15	14	8	2
9	3	22	16	10
17	11	5	4	23

On peut traduire cela avec des couleurs.
On tape :

```
lignec(n,k,C):={
local j,L;
L:=NULL;
pour j de 0 jusque n-1 faire
L:=L,affichage(carre(j+i*k,j+1+i*k),C[j]+rempli);
fpour;
return L;
}:;
carrec(n,C):={
local j,L,R;
R:=NULL;
pour j de 0 jusque n-1 faire
L:=lignec(n,n-j,C[j])
R:=R,L;
fpour;
return R;
}:;
```

Puis, `carrec(5,C)` renvoie :



Remarque
Si on affiche les nombres k modulo n (i.e. `irem(k,n)`), on obtient un carré magique latin (latin veut dire que chacun des entiers $0..n-1$ apparaissent dans chaque horizontale et verticale du carré et magique veut dire que la somme des éléments de chaque ligne, de chaque colonne et des 2 diagonales sont identiques et égales à $n(n^2-1)/2$). Pour l'exemple précédent, on tape :
`CR:=irem(cavalier(5,0,0),5)` et `carrec(5,CR)` on obtient :

0	4	3	2	1
3	2	1	0	4
1	0	4	3	2
4	3	2	1	0
2	1	0	4	3



On peut visualiser le déplacement du cavalier en mettant dans chaque case `iquo(k,n)` ce qui donne encore un carré latin. Pour l'exemple précédent, on tape :
`CQ:=(C-CQ)/5)` et `carrec(5,CQ)`, on obtient :
Pour l'exemple précédent, on obtient :

0	4	3	2	1
2	1	0	4	3
4	3	2	1	0
1	0	4	3	2
3	2	1	0	4



18.2.5 Produit de 2 carrés magiques

Pour construire des carrés eulériens pandiagonaux, il est souvent plus agéable d'écrire les coefficients du carré d'ordre n en base n. Tout entier compris entre 0 et n^2-1 s'écrit de manière unique sous la forme $n*a+b$ où a et b sont des entiers compris entre 0 et $n-1$.
Définition
On appelle produit AXB de 2 carrés magiques $A=a_{j,k}$ et $B=b_{j,k}$ de taille n vérifiant $a_{j,k}\in[0..n-1]$ et $b_{j,k}\in[0..n-1]$, le carré $C=c_{j,k}$ de taille n tel que : $c_{j,k}=na_{j,k}+b_{j,k}$ (i.e $a_{j,k},b_{j,k}$ est l'écriture en base n de $c_{j,k}$).
Donc, `C:=cavalier(5,0,0)` se décompose en le produit de 2 carrés latins pandiagonaux :

0	24	18	12	6		0	4	3	2	1		0	4	3	2	1
13	7	1	20	19		2	1	0	4	3		3	2	1	0	4
21	15	14	8	2	=	4	3	2	1	0	X	1	0	4	3	2
9	3	22	16	10		1	0	4	3	2		4	3	2	1	0
17	11	5	4	23		3	2	1	0	4		2	1	0	4	3

On peut traduire cela avec des couleurs.
On tape :

```
lignecc(n,k,C1,C2):={
local j,L;
```

```
L:=NULL;
pour j de 0 jusque n-1 faire
L:=L,affichage(rectangle(j+i*k,j+1/2+i*k,2),C1[j]+rempli),
     affichage(rectangle(j+1/2+i*k,j+1+i*k,2),C2[j]+rempli);
fpour;
return L;
}:;
carrecc(n,C1,C2):={
local j,L,R;
R:=NULL;
pour j de 0 jusque n-1 faire
L:=lignecc(n,n-j,C1[j],C2[j])
R:=R,L;
fpour;
return R;
}
:;
```

Puis :
`C:=cavalier(5,0,0);CR1:=iquo(C,5);CR2:=irem(C,5)`
`carrecc(5,CR1,CR2)`
renvoie :



18.2.6 Démonstration de l'algorithme du cavalier

Pour cela on va déterminer une expression générale de $A[j,k]$ en fonction de j et k lorsque l'on met $A[0,0]=0$ et que l'on applique l'algorithme du cavalier.
Pour cela on écrit $A[j,k]$ en base n :
$A[j,k]=na_{j,k}+b_{j,k}$ avec $a_{j,k}\in[0..n-1]$ et $b_{j,k}\in[0..n-1]$.
On a d'après l'algorithme du cavalier :
$a_{0,0}=0$ et $b_{0,0}=0$ $a_{j+1,k+2}=a_{j,k}$ et $b_{j+1,k+2}=b_{j,k}+1$ si $0\leq b_{j,k}<n-1$ et $a_{j,k-1}=a_{j,k}+1$ et $b_{j,k-1}=b_{j,k}+1=0 \bmod n$ si $b_{j,k}=n-1$
Le nombre $A[j,k]=na_{j,k}+b_{j,k}$ a donc comme indice de ligne $j=b_{j,k}-a_{j,k} \bmod n$.
En effet, quand $a_{j,k}=0$ on a $j=b_{j,k}$, puis quand $a_{j,k}=1$ et $b_{j,k}=0$ on a $j=n-1=b_{j,k}-a_{j,k} \bmod n$ etc...
Le nombre $A[j,k]=na_{j,k}+b_{j,k}$ a donc comme indice de colonne $k=2b_{j,k}-$

$3a_{j,k} \bmod n$.
En effet, quand $a_{j,k} = 0$ on a $k = 2b_{j,k} \bmod n$, donc on a bien $k = 2b_{j,k} - 3a_{j,k} \bmod n$
puis quand $a_{j,k} = 1$ et $b_{j,k} = 0$ on a $b_{j,k+1} = b_{j,k} - 1 \bmod n$ et $a_{j,k+1} = a_{j,k} - 1$ donc :
$b_{j,k} = b_{j,k+1} + 1 = 0 \bmod n$ et
$k+1 = 2b_{j,k+1} - 3a_{j,k+1} = 2b_{j,k} - 2 - 3a_{j,k} + 3 = 2b_{j,k} - 3a_{j,k} + 1 \bmod n$ (car $a_{j,k+1} = 0$ et d'après ce qui précéde si $a_{j,k+1} = 0$ on a $k+1 = 2b_{j,k+1} \bmod n$ et $b_{j,k-1} + 1 = 0 \bmod n$) donc $k = 2b_{j,k} - 3a_{j,k} \bmod n$ etc...
On a donc :
$j = b_{j,k} - a_{j,k} \bmod n$ et $k = 2b_{j,k} - 3a_{j,k} \bmod n$ donc
$2j - k = a_{j,k} \bmod n$ et $3j - k = b_{j,k} \bmod n$ donc
$a_{j,k} = 2j - k \bmod n$ et $b_{j,k} = 3j - k \bmod n$.

A est un carré eulérien pandiagonal.
En effet, les nombres $A[j,k]$ sont tous différents, car si $na_{j_1,k_1} + b_{j_1,k_1} = na_{j_2,k_2} + b_{j_2,k_2}$ alors
$a_{j_1,k_1} = a_{j_2,k_2}$ et $b_{j_1,k_1} = b_{j_2,k_2}$ donc
$2j_1 - k_1 = 2j_2 - k_2 \bmod n$ et $3j_1 - k_1 = 3j_2 - k_2 \bmod n$ donc
$k_2 - k_1 = 2j_2 - 2j_1 = 3j_2 - 3j_1$ donc $j_1 = j_2$ et $k_1 = k_2$.
Sur les $a_{j,k}$ et les $b_{j,k}$ forment un carré latin pandiagonal d'odre n car sur une même ligne les $a_{j,k}$ (respectivement les $b_{j,k}$) sont tous différents en effet :
si $a_{j_1,k_1} = a_{j_1,k_2}$ (resp $b_{j_1,k_1} = b_{j_1,k_2}$) alors
$2j_1 - k_1 = 2j_1 - k_2 \bmod n$ et $3j_1 - k_1 = 3j_1 - k_2 \bmod n$ donc $k_1 = k_2$.
Si n n'est pas un multiple de 2, ni un multiple de 3, sur une même colonne les $a_{j,k}$ (respectivement les $b_{j,k}$) sont tous différents en effet :
si $a_{j_1,k_1} = a_{j_2,k_1}$ (resp $b_{j_1,k_1} = b_{j_2,k_1}$) alors
$2j_1 - k_1 = 2j_2 - k_1 \bmod n$ (resp $3j_1 - k_1 = 3j_2 - k_1 \bmod n$)
donc $2(j_1 - j_2) = 0 \bmod n$ (resp $3(j_1 - j_2) = 0 \bmod n$).
comme 2 et 3 ne divisent pas n on en déduit que $j_1 = j_2$.
Sur les diagonales descendantes les $a_{j,k}$ (respectivement les $b_{j,k}$) sont tous différents en effet si :
$a_{j_1,k_1+j_1} = a_{j_2,k_1+j_2}$ (resp $b_{j_1,k_1+j_1} = b_{j_2,k_1+j_2}$) alors
$2j_1 - k_1 - j_1 = 2j_2 - k_1 - j_2 \bmod n$ et $3j_1 - k_1 - j_1 = 3j_2 - k_1 - j_2 \bmod n$
donc $j_1 = j_2$ (resp $j_1 = j_2$ car 2 ne divise pas n) Sur les diagonales montantes les $a_{j,k}$ (respectivement les $b_{j,k}$) sont tous différents en effet si :
$a_{j_1,k_1-j_1} = a_{j_2,k_1-j_2}$ (resp $b_{j_1,k_1-j_1} = b_{j_2,k_1-j_2}$) alors
$2j_1 - k_1 + j_1 = 2j_2 - k_1 + j_2 \bmod n$ et $3j_1 - k_1 + j_1 = 3j_2 - k_1 + j_2 \bmod n$ donc $j1 = j_2$ car 3 et 4 ne divisent pas n. Donc les carrés $a_{j,k}$ et $b_{j,k}$ sont des carrés latins pandiagonaux. Comme les nombres $A[j,k]$ sont tous différents, on en déduit que A est un carré eulérien pandiagonal.

18.2.7 Généralistion de l'algorithme du cavalier : $n \neq 2p$ et $n \neq 3p$

Lorsque $n = 5$ l'algorithme du cavalier a produit un carré eulérien pandiagonal qui était le produit de 2 carrés latins pandiagonaux. On va donc essayer de produire des carrés eulériens pandiagonaux comme produit de 2 carrés latins pandiagonaux.
Définition
On dit que p est un générateur interne de $\mathbb{Z}/n\mathbb{Z}$ si p, $p-1$ et $p+1$ sont des

générateurs de $\mathbb{Z}/n\mathbb{Z}$ (i.e p, $p-1$ et $p+1$ doivent être premier avec n).
Remarque si n est pair ou si n est un multiple de 3, $\mathbb{Z}/n\mathbb{Z}$ ne posséde pas de générateur interne.
Revenons à l'exemple obtenu avec la règle du cavalier pour $n = 5$, on a :
$C = A \times B =$

$$\begin{array}{|c|c|c|c|c|} \hline 0 & 24 & 18 & 12 & 6 \\ \hline 13 & 7 & 1 & 20 & 19 \\ \hline 21 & 15 & 14 & 8 & 2 \\ \hline 9 & 3 & 22 & 16 & 10 \\ \hline 17 & 11 & 5 & 4 & 23 \\ \hline \end{array} = \begin{array}{|c|c|c|c|c|} \hline 0 & 4 & 3 & 2 & 1 \\ \hline 2 & 1 & 0 & 4 & 3 \\ \hline 4 & 3 & 2 & 1 & 0 \\ \hline 1 & 0 & 4 & 3 & 2 \\ \hline 3 & 2 & 1 & 0 & 4 \\ \hline \end{array} \times \begin{array}{|c|c|c|c|c|} \hline 0 & 4 & 3 & 2 & 1 \\ \hline 3 & 2 & 1 & 0 & 4 \\ \hline 1 & 0 & 4 & 3 & 2 \\ \hline 4 & 3 & 2 & 1 & 0 \\ \hline 2 & 1 & 0 & 4 & 3 \\ \hline \end{array}$$

On remarque que :
le carré A est obtenu en mettant sur la première ligne, une permutation σ de $0..4$ (ici $\sigma([0, 1, 2, 3, 4]) = [0, 4, 3, 2, 1]$), puis la deuxième ligne est obtenue en décalant la première ligne vers la droite de 2 cases (i.e. $A[1, k) = A[0, k + 2]$ pour $k = 0..4$) ce qui donne [2,1,0,4,3] etc...
le carré B est obtenu en mettant sur la première ligne la même permutation de $0..4$ [0,4,3,2,1], puis la deuxième ligne est obtenue en décalant la première ligne vers la droite de 3 cases (i.e. $B[1, k) = A[0, k+3]$ pour $k = 0..4$) ce qui donne [3,2,1,0,4] etc...

On établit les résultats suivants :
Théorème1 La donnée de toute permutation σ de $0..n-1$ et de tout générateur p interne de $\mathbb{Z}/n\mathbb{Z}$ définit un carré latin de taille n.
Définition le carré latin ainsi construit est dit de type (σ, p).
Si A est un carré latin de taille n et de type (σ, p) on a :
$A[0, k) = \sigma([0, ..n-1])$ pour $k = 0..n-1$
$A[j + 1, k) = A[j, k + p \bmod n]$ pour $j = 0..n-1$ et pour $k = 0..n-1$
p engendre $\mathbb{Z}/n\mathbb{Z}$ donc $p \bmod n, p{+}1 \bmod n, p{+}n{-}1 \bmod n$ est une permutation de $[0, ..n-1]$.
Donc les lignes de A sont des permutations de $[0, ..n-1]$
Les colonnes de A sont telles que :
$A[j, 0] = A[j-1, p] = A[j-2, 2p \bmod n] = ..A[0, j * p \bmod n]$ et
pour pour j fixé et $k = 0..n-1$, on a :
$A[j, k] = A[j-1, p+k \bmod n] = A[j-2, 2p+k \bmod n] = .. =$
$A[0, j*p+k \bmod n]$
Donc les colonnes de A sont des permutations de $[0, 1..n-1]$
Les diagonales descendantes de A sont telles que pour k fixé et $j = 0..n-1$ on a :
$A[j, j + k \bmod n] = A[0, j*(p+1) + k \bmod n]$
$p{+}1$ engendre $\mathbb{Z}/n\mathbb{Z}$ donc les diagonales descendantes de A sont des permutations de $[0, ..n-1]$
Les diagonales montantes de A sont telles que pour k fixé et $j = 0..n-1$ on a :
$A[j, -j + k \bmod n] = A[0, j*(p-1) + k \bmod n]$
$p-1$ engendre $\mathbb{Z}/n\mathbb{Z}$ donc les diagonales montantes de A sont des permutations de $[0, ..n-1]$.
Théorème2 Le produit de 2 carrés latins de taille n et de types (σ, p) et (σ, q) est un carré eulérien pandiagonal si $p-q \bmod n$ est un générateur de $\mathbb{Z}/n\mathbb{Z}$.
En effet, soient A et B deux carrés d'ordre n où A est de type (σ, p) et B est de type (σ, q). (où σ est une bijection de $[0, 1..n-1]$ dans $[0, 1..n-1]$ et
$p-1 \bmod n$, p, $p+1 \bmod n$, $q-1 \bmod n$, q, $q-1 \bmod n$, $p-q \bmod n$ sont

des générateurs de $\mathbb{Z}/n\mathbb{Z}$).
On a par construction de A et B :
$\sigma(k) = A[0, k] = B[0, k]$
$A[1, k] = A[0, k - p \bmod n] = \sigma(k - p \bmod n)$ et
$A[j, k] = A[j, k - p * j \bmod n] = \sigma(k - p * j \bmod n)$
$B[1, k] = B[0, k - q \bmod n] = \sigma(k - q \bmod n)$ et
$B[j, k] = B[j, k - q * j \bmod n] = \sigma(k - q * j \bmod n)$
On pose $C[j, k] = n * A[j, k] + B[j, k]$ et montrons que C est un carré eulérien pandiagonal.
Chacun des entiers compris entre 0 et $n^2 - 1$ sont les coefficients de C. En effet les coefficients de C sont dans $[0, 1..n - 1]$ et ils sont tous différents car si :
$C[j_1, k_1] = n * A[j_1, k_1] + B[j_1, k_1] = C[j_2, k_2] = n * A[j_2, k_2] + B[j_2, k_2]$, on a :
$A[j_1, k_1] = A[j_2, k_2]$ et $B[j_1, k_1] = B[j_2, k_2]$ (unicité de l'écriture en base n) c'est à dire :
$\sigma(k_1 - p * j_1 \bmod n) = \sigma(k_2 - p * j_2 \bmod n)$ et
$\sigma(k_1 - q * j_1 \bmod n) = \sigma(k_2 - q * j_2 \bmod n) =$ et comme σ est une bijection on a :
$k_1 - k_2 = p*(j_1 - j_2) = q(j_1 - j_2) \bmod n$ donc $(p - q)(j_1 - j_2) = 0 \bmod n$ donc $j_1 = j_2$ (puisque $p - q \bmod n$ est un générateur de $\mathbb{Z}/n\mathbb{Z}$) et $k_1 = k_2$. Le carré C est pandiagonal car les $4n$ sommes des lignes, des colonnes, des diagonales de A et de B sont égales à $n(n - 1)/2$ (d'après le th 1) donc les sommes des lignes, des colonnes, des diagonales de C sont égales à $(n + 1)n(n - 1)/2 = n(n^2 - 1)/2$

18.2.8 Les programmes

Le produit de 2 carrés de $\mathbb{Z}/n\mathbb{Z}$

Voici le programme du produit de 2 carrés $A = a_{j,k}$ et $B = b_{j,k}$ de taille n vérifiant $a_{j,k} \in [0, 1..n - 1]$. On écrit le programme `estdansnz(A)` qui vérifie que les $A[j, k]$ sont dans $0, 1, ..n-1$ lorsque A est une matrice carrée de dimension n.
Puis, on écrit le programme `produitcar(A,B)` qui renvoie le produit de 2 carrés de dimension n dont les éléments sont dans $0, 1, ..n - 1$.
On tape :

```
estdansnz(A):={
local L,j,n1,rep;
L:=mat2list(A);
n1:=dim(L)-1;
rep:=vrai;
pour j de 0 jusque n1 faire
  si L[j]<0 ou L[j]>n1 alors
    retourne faux;
  fsi;
fpour;
retourne vrai;
}:;
produitcar(A,B):={
```

```
local j,k,C,sa,sb,n;
n:=size(A);
sa:=dim(A);
sb:=dim(B);
si sa!=[n,n] ou sb!=[n,n] ou estdansnz(A)==faux
   ou estdansnz(B)==faux alors
 retourne "erreur";
fsi;
C:=idn(n);
pour j de 0 jusque n-1 faire
pour k de 0 jusque n-1 faire
C[j,k]:=n*A[j,k]+B[j,k];
fpour;
fpour;
retourne C;
}:;
```

On tape :

`A:=[[0,4,3,2,1],[2,1,0,4,3],[4,3,2,1,0],[1,0,4,3,2],[3,2,1,0,4]]`
`B:=[[0,4,3,2,1],[3,2,1,0,4],[1,0,4,3,2],[4,3,2,1,0],[2,1,0,4,3]]`
`C:=produitcar(A,B)`

On obtient :

`[[0,24,18,12,6],[13,7,1,20,19],[21,15,14,8,2],[9,3,22,16,10],`
`[17,11,5,4,23]]`

On tape :

`estpandiage(C)`

On obtient :

`vrai`

Les permutations de $0..n$

Nous allons écrire un algorithme qui va renvoyer toutes les permutations de $l = [0..n-1]$ Les fonctions que l'on va écrire vont utiliser la fonction `echange`.

```
//echange ds l les elements d'indices j et k
echange(l,j,k):={
local a;
a:=l[j];
l[j]:=l[k];
l[k]:=a;
return l;
}:;
```

On peut décrire l'arbre des permutations de la liste $l[0..n-1]$:
à partir de la racine on a n=`size(l)` branches. Chaque branche commence respectivement par chacun des éléments de la liste l.
On va donc parcourir cet arbre de la racine (nœud de niveau 0) aux différentes extrémités, en renvoyant la liste des branches parcourues pour arriver à cette extrémité.
On va parcourir cet arbre en parcourant les n branches. On numérote ces n branches

par $p = 0..n-1$ et le niveau des nœuds $q = 0..n-1$.
On aura donc n appels récursifs.
Chaque branche p ($p = 0..n-1$) peut être considérée à leur tour comme un arbre ayant $n-1$ branches. La branche p aboutit aux permutations qui laissent invariant le p-ième élément de l ($l[p-1]$). C'est cet élément que l'on va échanger avec $l[0]$ pour que chaque branche p laisse invariant l' élément $l[0]$.
On sait que l'on est arrivé au bout de la branche, quand on se trouve au nœud de niveau $n-1$, dans ce cas la permutation chechée est l (c'est la permutation obtenue à partir de l en laissant ces $n-1$ premiers éléments invariants).
On utilise une variable locale `lr`, égale à la liste à renvoyer et un paramètre `k`, pour que `permuts(l,k)` renvoie toutes les permutations de `l` qui laissent invariant les k premiers éléments de `l`. On tape :

```
//permuts([1,2,3,4],0) utilise echange
permuts(l,k):={
local lr,j;
if (k==size(l)-1) return [l];
lr:=[];
for (j:=k;j<size(l);j++){
l:=echange(l,k,j);
lr:=[op(lr),op(permuts(l,k+1))];
}
return lr;
}:;
```

Remarque On n'est pas obligé de remettre la suite l à sa valeur de départ pour recommencer l'itération puisque le premier échange dans l'itération revient à transformer l en la liste où on a mis son j-ième élément en tête ($j = 0..n-1$).
Comme il faut 2 paramètres pour écrire la fonction récursive `permuts`, on écrit la fonction `permutation` qui utilise `permuts` :

```
//l:=[0,1,2,3];permutation(l);
//renvoie toutes les permutations de l
//utilise permuts
permutation(l):={
return permuts(l,0);
};
```

On tape :
`permutation([0,1,2])`
On obtient :
`[[0,1,2],[0,2,1],[1,0,2],[1,2,0],[2,0,1],[2,1,0]]`

Les générateurs internes de $\mathbb{Z}/n\mathbb{Z}$

Définition

On dit que $k\%n$ est un **générateur interne de** $\mathbb{Z}/n\mathbb{Z}$ si $k\%n$, $(k-1)\%n$, $(k+1)\%n$ sont des gérnérateurs de $\mathbb{Z}/n\mathbb{Z}$. Donc $k\%n$ est un générateur interne de $\mathbb{Z}/n\mathbb{Z}$ si $k, k-1, k+1$ sont premiers avec n.
Si n est un multiple de 2 ou 3 il n'y a donc pas de générateurs de $\mathbb{Z}/n\mathbb{Z}$.

Si n est premier les générateurs de $\mathbb{Z}/n\mathbb{Z}$ sont tous les entiers entre 2 et $n-2$:$2, 3..n-2$.
Pour faire le programme, on parcourt tous les entiers entre 1 et $n-1$. On compte combien d'entiers consécutifs sont premiers avec n. Ici c'est $j0, ..j-1$ le j étant celui qui a fait sortir du tantque :
`tantque j<=n-1 and d==1` (i.e. `j==n` ou `d=gcd(j,n)!=1`)
si ce nombre (qui est égal à $j-j0$) est supérieur ou égal à 3 on rajoute comme générateur `j0+1..j-2`.
puis on fait progresser j jusqu'à avoir `d=gcd(j,n)==1` ou `j==n` à la sortie du tantque :
`tantque j<=n-1 and d!=1.`
Puis on recommence...jusqu'à $j == n$

```
latin(P,p):={
local A,j,k;
n:=size(P);
A:=idn(n);
A[0]:=P;
pour j de 1 jusque n-1 faire
pour k de 0 jusque n-1 faire
A[j,k]:=P[irem(k-j*p,n)];
fpour;
fpour;
return A;
}:;
generateur(n):={
local L,d,j,j0;
L:=NULL;
si irem(n,2)==0 or irem(n,3)==0 then return L;fsi;
si isprime(n) alors return j$(j=2..n-2);fsi;
j:=1;
tantque j<=n-1 faire
j0:=j;
d:=gcd(j,n);
tantque j<=n-1 and d==1 faire
j:=j+1;
d:=gcd(j,n);
ftantque;
si j-j0>=3 alors
L:=L,j$(j=j0+1..j-2);
fsi;
j:=j+1;
d:=gcd(j,n);
tantque j<=n-1 and d!=1 faire
j:=j+1;
d:=gcd(j,n);
ftantque;
ftantque;
return L;
```

```
}:;
```

Fabrication d'un carré eulérien pandiagonal

On tape :

```
estgenerateur(p,n):={
local G;
p:=irem(p,n);
G:=generateur(n);
si member(p,G)==0 alors return faux;fsi
return vrai;
}:;
carrep(n):={
local G,P,L,j,k,A,B,p,q,s;
si irem(n,3)==0 ou irem(n,2)==0 alors renvoie "erreur2" fsi;
G:=generateur(n);
L:=[j$(j=0..n-1)];
P:=permutation(L);
s:=size(G);
k:=alea(s);
p:=G[k];
k:=alea(s);
q:=G[k];
tantque estgenerateur(p-q,n)==0 faire
k:=alea(s);
q:=G[k];
ftantque;
s:=size(P);
k:=alea(s);
P:=P[k];
A:=latin(P,p);
B:=latin(P,q);
retourne produitcar(A,B);
}
:;
```

Voici le regroupemenr de tous les programmes utilisés :

```
estpandiage(A):={
local j,k,n,s,L;
L:=mat2list(A);
si is_permu(L)==0 alors retourne faux fsi;
n:=size(A);
s:=n*(n^2-1)/2;
si [sum(A[j,k],j,0,n-1)$(k=0..n-1)]!=[s $(k=0..n-1)] alors
retourne faux;
fsi;
si [sum(A[k,j],j,0,n-1)$(k=0..n-2)]!=[s $(k=0..n-2)] alors
retourne faux;
```

```
fsi;
si [sum(A[j,irem(j+k,n)],j,0,n-1)$(k=0..n-2)]!=[s $(k=0..n-2)] alors
retourne faux;
fsi;
si [sum(A[j,irem(-j+k,n)],j,0,n-1)$(k=0..n-2)]!=[s $(k=0..n-2)] alors
retourne faux;
fsi;
retourne vrai;
}
:;
estdansnz(A):={
local L,j,n1,rep;
L:=mat2list(A);
n1:=dim(L)-1;
rep:=vrai;
pour j de 0 jusque n1 faire
si L[j]<0 ou L[j]>n1 alors retourne faux; fsi;
fpour;
retourne vrai;
}:;
produitcar(A,B):={
local j,k,C,sa,sb,n;
n:=size(A);
sa:=dim(A);
sb:=dim(B);
si sa!=[n,n] ou sb!=[n,n] ou estdansnz(A)==faux
   ou estdansnz(B)==faux alors
    retourne "erreur1";
fsi;
C:=idn(n);
pour j de 0 jusque n-1 faire
pour k de 0 jusque n-1 faire
C[j,k]:=n*A[j,k]+B[j,k];
fpour;
fpour;
retourne C;
}:;
echange(l,j,k):={
local a;
a:=l[j];
l[j]:=l[k];
l[k]:=a;
return l;
}:;
permuts(l,k):={
local lr,j;
if (k==size(l)-1) return [l];
lr:=[];
```

```
for (j:=k;j<size(l);j++){
l:=echange(l,k,j);
lr:=[op(lr),op(permuts(l,k+1))];
}
return lr;
}:;
permutation(l):={
return permuts(l,0);
}:;
latin(P,p):={
local A,j,k;
n:=size(P);
A:=idn(n);
A[0]:=P;
pour j de 1 jusque n-1 faire
pour k de 0 jusque n-1 faire
A[j,k]:=P[irem(k-j*p,n)];
fpour;
fpour;
return A;
}:;
generateur(n):={
local L,d,j,j0;
L:=NULL;
si irem(n,2)==0 or irem(n,3)==0 then return L;fsi;
si isprime(n) alors return j$(j=2..n-2);fsi;
j:=1;
tantque j<=n-1 faire
j0:=j;
d:=gcd(j,n);
tantque j<=n-1 and d==1 faire
j:=j+1;
d:=gcd(j,n);
ftantque;
si j-j0>=3 alors
L:=L,j$(j=j0+1..j-2);
fsi;
j:=j+1;
d:=gcd(j,n);
tantque j<=n-1 and d!=1 faire
j:=j+1;
d:=gcd(j,n);
ftantque;
ftantque;
return L;
}:;
estgenerateur(p,n):={
local G;
```

```
p:=irem(p,n);
G:=generateur(n);
si member(p,G)==0 alors return faux;fsi
return vrai;
}:;
carrep(n):={
local G,P,L,j,k,A,B,p,q,s;
si irem(n,3)==0 ou irem(n,2)==0 alors renvoie "erreur2" fsi;
G:=generateur(n);
L:=[j$(j=0..n-1)];
P:=permutation(L);
s:=size(G);
k:=alea(s);
p:=G[k];
k:=alea(s);
q:=G[k];
tantque estgenerateur(p-q,n)==0 faire
k:=alea(s);
q:=G[k];
ftantque;
s:=size(P);
k:=alea(s);
P:=P[k];
A:=latin(P,p);
B:=latin(P,q);
retourne produitcar(A,B);
}
:;
```

On tape :
`C:=carrep(7)`
On obtient :
```
[[16,48,24,8,0,40,32],[26,11,2,41,31,15,42],[1,35,33,18,44,27,10],
[34,17,43,21,12,4,37],[46,23,13,3,36,28,19],[7,5,39,30,20,45,22],
[38,29,14,47,25,9,6]]
```
On tape :
`estpandiage(C)`
On obtient `vrai`

18.2.9 Les carrés eulériens pandiagonaux d'ordre 2,3 et 4

Y-a-t-il des carres eulériens pandiagonaux d'ordre 2, d'ordre 3, d'ordre 4 ?
Rappel
Dans `Xcas` il y a l'instruction `is_permu` qui teste si une liste de dimension n est une permutation de $0..n-1$. Dans `Xcas` il y a l'instruction `mat2list` qui transforme une matrice en une liste. Ainsi `is_permu(mat2list(A))` nous dira si la matrice carrée `A` de dimension n est un carré eulérien d'ordre n i.e. chacun des nombres $0..n^2-1$ apparait dans les n^2 cases du carré.
Remarque

Si un carré d'ordre n est eulérien, pour qu'il soit pandiagonal il doit satisfaire à $4n$ équations. Mais comme la somme :
$\sum_{k=0}^{n^2-1} = n^2(n^2-1)/2$, la somme des éléments de chacune des $4n$ rangées (i.e lignes ou colonnes ou diagonales montantes ou descendantes) vaut donc $s = n(n^2-1)/2$. Il suffit donc de satisfaire à $4n-3$ équations (par exemple n équations pour les lignes, $n-1$ pour les colonnes, $n-1$ pour les diagonales montantes et $n-1$ pour les diagonales descendantes).

Cherchons si il y a des carrés magiques pandiagonaux d'ordre 2. Il y a 5 équations à satisfaire pour 4 inconnues.
0n tape :`A2:=[[x0,x1],[y0,y1]]; V2:=mat2list(A2)`
```
VS2:=linsolve([x0+x1-3,y0+y1-3,x0+y0-3,x1+y1-3,
x0+y1-3,x1+y0-3],V)
```
On obtient : `[3/2,3/2,3/2,3/2]`
Les 5 équations ne sont donc pas indépendantes.
Il n'y a donc pas de carrés magiques eulériens pandiagonaux d'ordre 2.
Cela était prévisible car la somme serait égale à 3 et donc on doit mettre 3 dans la ligne qui contient 0 et aussi 3 dans la colonne qui contient 0.

Cherchons si il y a des carrés magiques pandiagonaux d'ordre 3. Il y a 9 équations à satisfaire pour 9 inconnues.
0n tape :
```
A3:=[[x0,x1,x2],[y0,y1,y2],[z0,z1,z2]];V3:=mat2list(A3);
VS3:=linsolve([x0+x1+x2+-12,y0+y1+y2-12,z0+z1+z2-12,
x0+y0+z0-12,x1+y1+z1-12,x2+y2+z2-12,x0+y1+z2-12,x1+y2+z0-12,
x2+y0+z1-12,x2+y1+z0-12,x0+y2+z1-12,x1+y0+z2-12],V3)
```
On obtient : `[4,4,4,4,4,4,4,4,4]`
Il n'y a donc pas de carrés magiques pandiagonaux d'ordre 3.
Cela était encore prévisible car la somme serait égale à 12. On a :
8+3+1=8+4+0=12 donc on ne peut pas mettre des nombres différents, dans la colonne du 8, dans la ligne du 8 et dans les diagonales contenant 8.

Cherchons si il y a des carrés magiques pandiagonaux d'ordre 4. 0n tape :
```
A4:=[[x0,x1,x2,x3],[y0,y1,y2,y3],[z0,z1,z2,z3],[t0,t1,t2,t3]];
V4:=mat2list(A4)
```
Il y a 13 équations à satisfaire pour 16 inconnues ($s = 30$). On a donc nos chances !
```
linsolve([x0+x1+x2+x3-30,y0+y1+y2+y3-30,z0+z1+z2+z3-30,
t0+t1+t2+t3-30,x0+y0+z0+t0-30,x1+y1+z1+t1-30,x2+y2+z2+t2-30,
x3+y3+z3+t3-30,x0+y1+z2+t3-30,x1+y2+z3+t0-30,x2+y3+z0+t1-30,
x3+y0+z1+t2-30,x0+y3+z2+t1-30,x1+y0+z3+t2-30,x2+y1+z0+t3-30,
x3+y2+z1+t0-30],V4)
```
On obtient : `[t2+t3+z3-15,-z3+15,-t1-t2+z3+15,t1-t3-z3+15,-t2+15,-t3+15,t1+t2+t3-15,-t1+15,t1+t2-z3,-t1+t3+z3,-t2-t3-z3+30,z3,-t1-t2-t3+30,t1,t2,t3]` On en déduit que les 13 équations ne sont pas indépendantes : On remarque que si on fait la sommes des 6 rangées ci-dessous la sommes des cases notées par des points rouges est comptés 3 fois donc ces 4 cases ont aussi comme somme s ce qui entraine que la somme des 4 cases situées aux sommets vaut aussi s.



On en déduit que la somme des cases A, B, C, D, E, F, G, H vaut $2s$, c'est à dire que si la somme des cases A, B, C, D vaut s cela entraine que la somme des cases E, F, G, H vaut aussi s.

On tape :

```
L4(z3,t1,t2,t3):=[t2+t3+z3-15,-z3+15,-t1-t2+z3+15,t1-t3-z3+15,
-t2+15,-t3+15,t1+t2+t3-15,-t1+15,t1+t2-z3,-t1+t3+z3,-t2-t3-z3+30,
z3,-t1-t2-t3+30,t1,t2,t3]
```

puis on va chercher avec un programme les valeurs de $z3, t1, t2, t3$ pour lesquelles on a des carrés eulériens. Pour cela on va faire varier les 4 variables $z3, t1, t2, t3$ entre 0 et 15.

On tape :

```
carremp4(L):={
local R,z3,t1,t2,t3;
R:=[];
pour z3 de 0 jusque 15 faire
 pour t1 de 0 jusque 15 faire
  pour t2 de 0 jusque 15 faire
   pour t3 de 0 jusque 15 faire
    //si listeab(L(z3,t1,t2,t3),0,15) alors
    si is_permu(L(z3,t1,t2,t3)) alors
       R:=append(R,[z3,t1,t2,t3]);
    fsi
   fpour;
  fpour;
 fpour;
fpour;
return R;
}:;
```

On tape :

`LS4:=carremp4(L4):;`

Après "Evaluation time : 15.58", on obtient "Done"

On tape : `size(LS4)` et on obtient `384`

On tape : `list2mat(L4(op(LS4[0])),4)`

On obtient `[[2,15,8,5],[9,4,3,14],[7,10,13,0],[12,1,6,11]]`

Si on échange 2 lignes (ou 2 colonnes) dans un carré pandiagonal, on obtient encore un carré pandiagonal : il y a donc 384/16=24 carrés pandiagonaux vraiment différents. On peut donc améliorer le programme précédent en supposant que $t3 = 15$.

On tape :
`L41(z3,t1,t2):=L4(z3,t1,t2,15)`

```
carremp41(L):={
local R,z3,t1,t2;
R:=[];
pour z3 de 0 jusque 14 faire
pour t1 de 0 jusque 14 faire
pour t2 de 0 jusque 14 faire
si is_permu(L(z3,t1,t2)) alors R:=append(R,[z3,t1,t2]);fsi;
//si listeab(L(z3,t1,t2),0,15) alors R:=append(R,[z3,t1,t2]);fsi;
fpour;
fpour;
fpour;
return R;
}:;
```

On tape :
`LS41:=carremp41(L41):;`
Après "Evaluation time : 1.23" on obtient "Done"
`size(LS41)` renvoie `24` et
`list2mat(L41(op(LS41[0])),4)` renvoie
`[[5,14,9,2],[11,0,7,12],[6,13,10,1],[8,3,4,15]]`
Soit :
`C:=[[5,14,9,2],[11,0,7,12],[6,13,10,1],[8,3,4,15]]`
On écrit `C` en base 4 on obtient :
`[[11,32,21,2],[23,0,13,30],[12,31,22,1],[20,3,10,33]]`

c'est à dire :

$$C = A \text{ X } B = \begin{array}{|c|c|c|c|}\hline 5 & 14 & 9 & 2 \\ \hline 11 & 0 & 7 & 12 \\ \hline 6 & 13 & 10 & 1 \\ \hline 8 & 3 & 4 & 15 \\ \hline\end{array} = \begin{array}{|c|c|c|c|}\hline 1 & 3 & 2 & 0 \\ \hline 2 & 0 & 1 & 3 \\ \hline 1 & 3 & 2 & 0 \\ \hline 2 & 0 & 1 & 3 \\ \hline\end{array} \text{ X } \begin{array}{|c|c|c|c|}\hline 1 & 2 & 1 & 2 \\ \hline 3 & 0 & 3 & 0 \\ \hline 2 & 1 & 2 & 1 \\ \hline 0 & 3 & 0 & 3 \\ \hline\end{array}$$

)=
On tape :
`A:=[[1,3,2,0],[2,0,1,3],[1,3,2,0],[2,0,1,3]]`
`B:=[[1,2,1,2],[3,0,3,0],[2,1,2,1],[0,3,0,3]]`
`C:=produitcar(A,B)`
On obtient :
`[[5,14,9,2],[11,0,7,12],[6,13,10,1],[8,3,4,15]]`
On tape :
`estpandiage(C)`
On obtient :
`vrai`
Remarque Pour cet exemple ($n = 4$), dans la décomposition de C en produit AXB les carrés A et B ne sont pas latins.
Cherchons les 24 carrés d'ordre 4 se terminant par 0
On tape :
`LL41(z3,t1,t2):=L4(z3,t1,t2,0)`

```
LLS41:=carremp41(L41):;
size(LLS41)
```
On obtient :
```
24
```
On tape :
```
list2mat(LL41(op(LLS41[1])),4)$(k=0..23)
```
On obtient :

```
[[5,8,6,11],[2,15,1,12],[9,4,10,7],[14,3,13,0]],
[[6,8,5,11],[1,15,2,12],[10,4,9,7],[13,3,14,0]],
[[3,8,6,13],[4,15,1,10],[9,2,12,7],[14,5,11,0]],
[[6,8,3,13],[1,15,4,10],[12,2,9,7],[11,5,14,0]],
[[3,8,5,14],[4,15,2,9],[10,1,12,7],[13,6,11,0]],
[[5,8,3,14],[2,15,4,9],[12,1,10,7],[11,6,13,0]],
[[9,4,10,7],[2,15,1,12],[5,8,6,11],[14,3,13,0]],
[[10,4,9,7],[1,15,2,12],[6,8,5,11],[13,3,14,0]],
[[3,4,10,13],[8,15,1,6],[5,2,12,11],[14,9,7,0]],
[[10,4,3,13],[1,15,8,6],[12,2,5,11],[7,9,14,0]],
[[3,4,9,14],[8,15,2,5],[6,1,12,11],[13,10,7,0]],
[[9,4,3,14],[2,15,8,5],[12,1,6,11],[7,10,13,0]],
[[9,2,12,7],[4,15,1,10],[3,8,6,13],[14,5,11,0]],
[[12,2,9,7],[1,15,4,10],[6,8,3,13],[11,5,14,0]],
[[5,2,12,11],[8,15,1,6],[3,4,10,13],[14,9,7,0]],
[[12,2,5,11],[1,15,8,6],[10,4,3,13],[7,9,14,0]],
[[5,2,9,14],[8,15,4,3],[6,1,10,13],[11,12,7,0]],
[[9,2,5,14],[4,15,8,3],[10,1,6,13],[7,12,11,0]],
[[10,1,12,7],[4,15,2,9],[3,8,5,14],[13,6,11,0]],
[[12,1,10,7],[2,15,4,9],[5,8,3,14],[11,6,13,0]],
[[6,1,12,11],[8,15,2,5],[3,4,9,14],[13,10,7,0]],
[[12,1,6,11],[2,15,8,5],[9,4,3,14],[7,10,13,0]],
[[6,1,10,13],[8,15,4,3],[5,2,9,14],[11,12,7,0]],
[[10,1,6,13],[4,15,8,3],[9,2,5,14],[7,12,11,0]]
```

Dans la gravure Melancolia I d'Albrecht Dürer il y a un carré magique en haut et a droite qui est eulérien (écrit avec les nombres de 1 à 16) mais il n'est pas pandiagonal. Il peut être représenté (avec les nombres de 0 à 15) par :
$[[15, 2, 1, 12], [4, 9, 10, 7], [8, 5, 6, 11], [3, 14, 13, 0]]$
Ce carré a sur ces lignes les mêmes nombres que sur les lignes de ces 3 carrés pandiagonaux :

```
[[5,8,6,11],[2,15,1,12],[9,4,10,7],[14,3,13,0]],
[[9,4,10,7],[2,15,1,12],[5,8,6,11],[14,3,13,0]],
[[10,4,9,7],[1,15,2,12],[6,8,5,11],[13,3,14,0]]
```

18.2.10 Existence de carrés eulériens pandiagonaux d'ordre 6

Y-a-t-il des carres eulériens pandiagonaux d'ordre 6 ? On tape :
```
V6:=[x0,x1,x2,x3,x4,x5,y0,y1,y2,y3,y4,y5,z0,z1,z2,z3,z4,z5,
```

```
t0,t1,t2,t3,t4,t5,u0,u1,u2,u3,u4,u5,v0,v1,v2,v3,v4,v5]
M:=list2mat(V6,6)
L(j):=sum(M[j,k],k=0..5)
C(k):=sum(M[j,k],j=0..5)
D(k):=sum(M[j,irem(-j+k,6)],j=0..5)
T(k):=sum(M[j,irem(k+j,6)],j=0..5)
LL:=(L(j)-105)$(j=0..5)
CC:=(C(k)-105)$(k=0..4)
DD:=(D(k)-105)$(k=0..4)
TT:=(T(k)-105)$(k=0..4)
linsolve([LL,CC,DD,TT],V6)[0]
```

On obtient la valeur de `x0` :
`t4+t5+u3+u4+u5+v2+v3+v4+v5+z5+(-315)/2`
Donc `x0` n'est jamais un nombre entier.
Il n'y a donc pas de carrés eulériens pandiagonaux d'ordre 6.

18.3 Les hexagones magiques

18.3.1 Le dessin

Peut-on placer les entiers de 1 à n dans une structure hexagonale ayant n hexagones, de façon à ce que la somme des nombres alignés soit constante ?

Pour une structure hexagonale ayant 7 hexagones :



On voit facilement que cela est impossible puisque si $x1 + x3 = x3 + x6$ alors $x1 = x6$

Pour une structure hexagonale ayant 19 hexagones, il faut que les nombres 1,2..19 remplissent les 19 cases afin que la somme de toutes les cases alignèes soit égale à 38 (en tout 5*3 équations à satisfaire). Si il y a yne solution, puisque 1+2+...19=190 et qu'il y a en tout 5 rangées disjointes de même somme, la somme de chaque rangèe vaut 190/5=38.



On choisit le nom des variables :



On tape :

```
linsolve([x1+x2+x3=38,x4+x5+x6+x7=38,x8+x9+x10+x11+x12=38,
x13+x14+x15+x16=38,x17+x18+x19=38,x1+x4+x8=38,x2+x5+x9+x13=38,
x3+x6+x10+x14+x17=38,x7+x11+x15+x18=38,x12+x16+x19=38,
x3+x7+x12=38,x2+x6+x11+x16=38,x1+x5+x10+x15+x19=38,
x4+x9+x14+x18=38,x8+x13+x17=38],
[x1,x2,x3,x4,x5,x6,x7,x8,x9,x10,
x11,x12,x13,x14,x15,x16,x17,x18,x19])
```

On obtient :

```
[-x10-x11-x14-2*x15-x16-x18-x19+76,x10+x14+x15,
x11+x15+x16+x18+x19-38,x10+x11+x15,x11+x14+x15+x16+x18-38,
-x10-x11-x14-x15-x16+38,-x11-x15-x18+38,x14+x15+x16+x18+x19-38,
-x10-x11-x14-x15-x18+38,x10,x11,-x16-x19+38,
-x14-x15-x16+38,x14,x15,x16,-x18-x19+38,x18,x19]
```

On voit que les solutions réelles dépendent des 7 variables :
$x10, x11, x14, x15, x16, x18, x19$.

18.3.2 Le programme

On va donc écrire un programme pour chercher les solutions qui sont des permutations de [1,2,...19].

Pour éviter de faire $19^7 = 893871739$ itérations on remarque que :
les variables $x2$ et $x4$ sont indépendantes de $x16, x18, x19$:
les variables $x6$ et $x13$ sont indépendantes de $x18, x19$:
les variables $x5$, $x7$ et $x9$,$x2$ et $x4$ sont indépendantes de $x19$
On peut remarquer de plus que : $x1$ (resp $x3, x8, x12, x17, x19$) ne peut pas être égal à 1 ou à 2 car $x1 + x2 + x3 = x1 + x4 + x8 = 38$:
si $x1 = 1$ cela entraine $x2 + x3 = x4 + x8 = 37$. Mais l'équation $x + y = 37$ a comme solution dans 1...19 que les couples (19,18) et (18,19) ce qui ne donne pas 4 solutions distinctes pour $x2, x3, x4, x8$.
si $x1 = 2$ cela entraine $x2 + x3 = x4 + x8 = 36$. Mais l'équation $x + y = 36$ a comme solution dans 1...19 les couples (19,17), (18,18) et (17,19) ce qui ne donne pas 4 solutions distinctes pour $x2, x3, x4, x8$.
On écrit sans trop chercher à optimiser :

```
hexamagique():={
local x1,x2,x3,x4,x5,x6,x7,x8,x9,x10,x11,x12,
      x13,x14,x15,x16,x17,x18,x19,L,nsol,nsol2;
L:=NULL;
nsol:=[1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19];
nsol2:=[3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19];
pour x10 de 1 jusque 19 faire
pour x11 de 1 jusque 19 faire
pour x14 de 1 jusque 19 faire
pour x15 de 1 jusque 19 faire
 x2:=x10+x14+x15; x4:=x10+x11+x15;
 si member(x2,nsol) and member(x4,nsol) and x2!=x4 alors
  pour x16 de 1 jusque 19 faire
   x13:=-x14-x15-x16+38;
   x6:=-x10-x11+x13;//x6:=-x10-x11-x14-x15-x16+38
   si member(x13,nsol) and member(x6,nsol) and x13!=x6  alors
    pour x18 de 1 jusque 19 faire
     x7:=-x11-x15-x18+38;
     x5:=-x7+x14+x16;//x5:=x11+x14+x15+x16+x18-38;
     x9:=-x10+x7-x14;//x9:=-x10-x11-x14-x15-x18+38;
     si member(x5,nsol) and member(x7,nsol) and member(x9,nsol)
        and x5!=x7  and x5!=x9 and x9!=x7 alors
      pour x19 de 3 jusque 19 faire
       x17:=-x18-x19+38; x12:=-x16-x19+38;
       si member(x17,nsol2) and member(x12,nsol2) and
        x12!=x17 alors
        //x1:=-x10-x11-x14-2*x15-x16-x18-x19+76;
        x1:=-x10-x11-x14-2*x15-x16+x17+38;
        x8:=x14+x15+x16-x17;//x8:=x14+x15+x16+x18+x19-38;
        x3:=x11+x15+x16-x17;//x3:=x11+x15+x16+x18+x19-38;
        si is_permu([x1,x2,x3,x4,x5,x6,x7,x8,x9,x10,x11,x12,
               x13,x14,x15,x16,x17,x18,x19]-[1$19]) alors
         L:=L,[x1,x2,x3,x4,x5,x6,x7,x8,x9,x10,x11,x12,
             x13,x14,x15,x16,x17,x18,x19];
        fsi;
```

```
        fsi;
       fpour;
      fsi;
     fpour;
    fsi;
   fpour;
  fsi;
fpour;
fpour;
fpour;
fpour;
return L
}:;
```

On tape :
`Rep:=hexamagique()`
On obtient (Evaluation time : 1011.94 c'est long ! mais l'employé de chemin de fer, Clifford W. Adams a mis 47 ans de 1910 à 1957 pour trouver une solution avec des hexagones en céramique numérotés de 1 à 19...) :

```
[15,14,9,13,8,6,11,10,4,5,1,18,12,2,7,17,16,19,3],
[16,19,3,12,2,7,17,10,4,5,1,18,13,8,6,11,15,14,9],
[15,13,10,14,8,4,12,9,6,5,2,16,11,1,7,19,18,17,3],
[18,17,3,11,1,7,19,9,6,5,2,16,14,8,4,12,15,13,10],
[3,19,16,17,7,2,12,18,1,5,4,10,11,6,8,13,9,14,15],
[9,14,15,11,6,8,13,18,1,5,4,10,17,7,2,12,3,19,16],
[3,17,18,19,7,1,11,16,2,5,6,9,12,4,8,14,10,13,15],
[10,13,15,12,4,8,14,16,2,5,6,9,19,7,1,11,3,17,18],
[9,11,18,14,6,1,17,15,8,5,7,3,13,4,2,19,10,12,16],
[10,12,16,13,4,2,19,15,8,5,7,3,14,6,1,17,9,11,18],
[16,12,10,19,2,4,13,3,7,5,8,15,17,1,6,14,18,11,9],
[18,11,9,17,1,6,14,3,7,5,8,15,19,2,4,13,16,12,10]
```

Ces 12 solutions représentent en faite la même solution puisque une solution donne 5 solutions par rotation de $k * \pi/3$ pour $(k = 1..5)$ et donne 6 solutions par symétrie par rapport aux droites passant par le centre de d'angle $k * \pi/3$ pour $(k = 0..5)$.
Voici la première solution :



On peut remarquer que l'on peut éliminer les solutions obtenues par rotation en faisant un programme qui impose successivement, par exemple, $x11 = 1$, puis $x16 = 1$. Mais si $x16 = 1$ alors $x12 + x16 + x19 = 38$ entraine $x12 + x19 = 37$ donc seuls le couples (18,19) et (19,18) sont solutions. Ces 2 solutions donneront des résultas symétriques, donc on va imposer successivement $x11 = 1$, puis $x16 = 1,\ x12 = 18,\ x19 = 19$. On aura ainsi les solutions symétriques pour $x11 = 1$ mais pas pour $x16 = 1$
On va aussi, pour réduire le temps d'exécution, faire un test, à chaque étape, qui vérifie que les valeurs obtenues sont différentes et comprises entre 1 et 19.
Pour cela, on tape `tousdiff(L,LR)` qui teste si les éléments de `L` sont tous différents et sont dans la liste `LR` :

```
tousdiff(L,LR):={
local s,a;
s:=size(L);
tantque s!=0 faire
a:=L[0];
si member(a,LR) alors
  L:=remove(a,L);
  si s!=size(L)+1 alors return 0; fsi;
  s:=s-1;
sinon return 0;
fsi
ftantque;
return 1;
}:;
```

puis on utilise ce test dans `hexamagique2()` :

```
hexamagique2():={
local x1,x2,x3,x4,x5,x6,x7,x8,x9,x10,x11,x12,LR,
      x13,x14,x15,x16,x17,x18,x19,L,nsol,nsol2;
LR:=NULL;
nsol:=[1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19];
x11:=1;
pour x10 de 2 jusque 19 faire
pour x14 de 2 jusque 19 faire
pour x15 de 2 jusque 19 faire
 x2:=x10+x14+x15; x4:=x10+x11+x15;
 L:=[x2,x4,x10,1,x14,x15];
 si tousdiff(L,nsol) alors
  pour x16 de 2 jusque 19 faire
   x13:=-x14-x15-x16+38; x6:=-x10-x11+x13;
   L:=[x2,x4,x6,x10,x11,x13,x14,x15,x16];
   si  tousdiff(L,nsol) alors
    pour x18 de 2 jusque 19 faire
     x7:=-x11-x15-x18+38; x5:=-x7+x14+x16; x9:=-x10+x7-x14;
     L:=[x2,x4,x5,x6,x7,x9,x10,1,x13,x14,x15,x16,x18];
```

```
     si  tousdiff(L,nsol) alors
      pour x19 de 3 jusque 19 faire
       x17:=-x18-x19+38; x12:=-x16-x19+38;
       x1:=-x10-x11-x14-2*x15-x16+x17+38;
       x8:=x14+x15+x16-x17; x3:=x11+x15+x16-x17;
       L:=[x1,x2,x3,x4,x5,x6,x7,x8,x9,x10,1,x12,
              x13,x14,x15,x16,x17,x18,x19];
       si tousdiff(L,nsol) alors
        LR:=LR,L;
       fsi;
      fpour;
     fsi;
    fpour;
   fsi;
  fpour;
 fsi;
fpour;
fpour;
fpour;
x16:=1;x12:=18;x19:=19;
pour x10 de 2 jusque 17 faire
pour x11 de 2 jusque 17 faire
pour x14 de 2 jusque 17 faire
pour x15 de 2 jusque 17 faire
 x2:=x10+x14+x15; x4:=x10+x11+x15;
  x13:=-x14-x15+37;
  x6:=-x10-x11+x13;
  L:=[x2,x4,x6,x10,x11,18,x14,x15,1,x18,19];
  si tousdiff(L,nsol) alors
   pour x18 de 2 jusque 17 faire
    x7:=-x11-x15-x18+38;
    x5:=-x7+x14+x16;
    x9:=-x10+x7-x14;
    x17:=-x18-x19+38;
    x1:=-x10-x11-x14-2*x15-x16+x17+38;
    x8:=x14+x15+x16-x17;
    x3:=x11+x15+x16-x17;
    L:=[x1,x2,x3,x4,x5,x6,x7,x8,x9,x10,x11,18,
              x13,x14,x15,1,x17,x18,19];
    si  tousdiff(L,nsol) alors
      LR:=LR,L;
    fsi;
   fpour;
  fsi;
fpour;
fpour;
fpour;
fpour;
```

```
return LR;
}
:;
```

On tape :
`Rep:=hexamagique2()`
On obtient la première solution trouvée tout à l'heure et sa symétrique (Evaluation time : 25.39 on a divisé le temps du premier programme presque par 40 mais on peut certainement faire encore mieux ! ! ! !) :
`[15,14,9,13,8,6,11,10,4,5,1,18,12,2,7,17,16,19,3],`
`[16,19,3,12,2,7,17,10,4,5,1,18,13,8,6,11,15,14,9]`

18.3.3 Les programmes des figures

Pour faire le dessin on tape :

```
hexag2():={
local L,j;
L:=hexagone(-1/2-i*sqrt(3)/2,1/2-i*sqrt(3)/2);
pour j de 0 jusque 5 faire
 L:=L,hexagone((1/2+i*sqrt(3)/2)*exp(i*j*pi/3),exp(i*j*pi/3));
fpour;
L:=L,legende(i*1.73,"x3",red);
L:=L,legende(-1.5+i*0.86,"x1",red);
L:=L,legende(1.5+i*0.86,"x6",red);
retourne L;
}

:;
hexag():={
local L,j;
L:=hexagone(-1/2-i*sqrt(3)/2,1/2-i*sqrt(3)/2);
pour j de 0 jusque 5 faire
 L:=L,hexagone((1/2+i*sqrt(3)/2)*exp(i*j*pi/3),exp(i*j*pi/3)),
hexagone((5/2+i*sqrt(3)/2)*exp(i*j*pi/3),2*exp(i*j*pi/3)),
hexagone((2+i*sqrt(3))*exp(i*j*pi/3),(5/2+i*sqrt(3)/2)*exp(i*j*pi/3));
fpour;
retourne L;
}
:;
hexag2():={
local L,j;
L:=hexagone(-1/2-i*sqrt(3)/2,1/2-i*sqrt(3)/2);
pour j de 0 jusque 5 faire
 L:=L,hexagone((1/2+i*sqrt(3)/2)*exp(i*j*pi/3),exp(i*j*pi/3)),
fpour;
L:=legende(-3.3+i*1.73,"x1",red);
L:=legende(-3.3+i*1.73,"x1",red);
L:=legende(-3.3+i*1.73,"x1",red);
```

```
retourne L;
}
:;
leg():={
local L;
L:=legende(-3.3+i*1.73,"x1",red);
L:=L,legende(-3.3,"x2",red);
L:=L,legende(-3.3-i*1.73,"x3",red);
L:=L,legende(-1.8+2.5*i,"x4",red);
L:=L,legende(-1.8+0.86*i,"x5",red);
L:=L,legende(-1.8-0.86*i,"x6",red);
L:=L,legende(-1.8-2.5*i,"x7",red);
L:=L,legende(2.7+i*1.73,"x17",red);
L:=L,legende(2.7,"x18",red);
L:=L,legende(2.7-i*1.73,"x19",red);
L:=L,legende(-0.3+i*3.5,"x8",red);
L:=L,legende(-0.3+i*1.73,"x9",red);
L:=L,legende(-0.3,"x10",red);
L:=L,legende(-0.3-i*1.73,"x11",red);
L:=L,legende(-0.3-i*3.5,"x12",red);
L:=L,legende(1.2+2.5*i,"x13",red);
L:=L,legende(1.2+0.86*i,"x14",red);
L:=L,legende(1.2-0.86*i,"x15",red);
L:=L,legende(1.2-2.5*i,"x16",red);
L;}
:;
valleg(Rep):={
local L,x1;
[x1,x2,x3,x4,x5,x6,x7,x8,x9,x10,x11,x12,x13,x14,
 x15,x16,x17,x18,x19]:=Rep[0];
L:=legende(-3.3+i*1.73,string(x1));
L:=L,legende(-3.3,string(x2));
L:=L,legende(-3.3-i*1.73,string(x3));
L:=L,legende(-1.8+2.5*i,string(x4));
L:=L,legende(-1.8+0.86*i,string(x5));
L:=L,legende(-1.8-0.86*i,string(x6));
L:=L,legende(-1.8-2.5*i,string(x7));
L:=L,legende(2.7+i*1.73,string(x17));
L:=L,legende(2.7,string(x18));
L:=L,legende(2.7-i*1.73,string(x19));
L:=L,legende(-0.3+i*3.5,string(x8));
L:=L,legende(-0.3+i*1.73,string(x9));
L:=L,legende(-0.3,string(x10));
L:=L,legende(-0.3-i*1.73,string(x11));
L:=L,legende(-0.3-i*3.5,string(x12));
L:=L,legende(1.2+2.5*i,string(x13));
L:=L,legende(1.2+0.86*i,string(x14));
L:=L,legende(1.2-0.86*i,string(x15));
```

```
L:=L,legende(1.2-2.5*i,string(x16));
L;}:;
```

Chapitre 19

Les jeux

19.1 Un jeu

On considère le jeu suivant :

— Le joueur tire au hasard un nombre entre 1 et 10. Ce nombre est son gain, il peut le garder et la partie est finie.

— Le joueur peut choisir d'abandonner ce gain et de recommencer. Il lui est permis de tirer ainsi au plus 5 fois et son gain est le résultat du dernier tirage lorsqu'il a abandonné les gains des 4 tirages précédents.

1. Écrire un programme qui réalise ce jeu,
2. Le joueur choisit 4 nombres n_1, n_2, n_3, n_4 compris entre 1 et 10 et choisit comme stratégie de garder son gain du $j$ième essai si celui-ci est supérieur ou égal à n_j sinon il recommence (si n_1=9, n_2=8, n_3=6, n_4=5 avec le tirage 2,7,8, le gain est 8 mais avec le tirage 8,7,5,4,2 le gain est 2).
   Il joue 100 parties selon cette stratégie avec les mêmes n_j. Écrire un programme donnant la moyenne des gains obtenus. A vous de jouer !.

```
jeu1():={
 //repondre o pour dire oui
 local j,g,rep;
 pour j de 1 jusque 4 faire
   g:=alea(10)+1;afficher(g);
   lis_phrase(rep);
   si rep=="o" alors retourne g; fsi;
 fpour;
 retourne alea(10)+1;
}:;
```

`jeu2` a comme paramètre `L` qui est la séquence n_1, n_2, n_3, n_4 puis LL est la suite $n_1, n_2, n_3, n_4, 0$ de façon à ce que $LL[j]$ soit définie pour $j = 0..4$ ($LL[0] = n_1$). `jeu102` est la moyenne des gains de 100 parties faites avec `jeu2`.

```
jeu2(L):={
 local j,g,LL;
 LL:=L,0;
 pour j de 0 jusque 4 faire
```

```
   g:=alea(10)+1;afficher(g);
   si g>=LL[j] alors break fsi;
 fpour;
 retourne g;
}:;
jeu102(L):={
 local g,j;
 g:=0;
 pour j de 1 jusque 100 faire
   g:=g+jeu2(L);
 fpour;
 retourne evalf(g/100);
}:;
```

À vous de trouver la meilleure stratégie !

19.2 Le Master-Mind

19.2.1 La règle du jeu

L'ordinateur sera ici le codificateur : il choisit une combinaison de 4 couleurs, différentes ou non, parmi les 6 disponibles numérotées de 0 à 5 (noir, rouge, vert, jaune, bleu, magenta). Par exemple (0,5,2,5)
Le joueur doit essayer de deviner la combinaison choisit en donnant une proposition. Par exemple (5,4,2,1).
Le codificateur donne alors son verdict en donnant 1 marqueur noir par couleur se trouvant à la bonne place et un marqueur blanc par couleur trouvée mais ne se trouvant pas à la bonne place. Avec l'exemple il répond 1N et 1B.
Le joueur peut faire au plus nc propositions (en général $nc = 10$).
Variante On peut permettre une septième couleur de numéro 6 (cyan) mais le codificateur ne peut pas choisir la combinaison (6,6,6,6).

19.2.2 Le programme du Master-Mind

```
masterm(nc):={
local L,R,R0,n,j,N,B,a;
L:=[alea(6)$(j=1..4)];
n:=0;
while (n<nc) {
saisir(R);
R:=[R];n:=n+1;R0:=R;
N:=0;B:=0;j:=0;
pour j de 0 jusque 3 faire
a:=R[j];si a==L[j] alors N:=N+1; fsi;
fpour;
if (N==4) alors return "gagne en "+n+" coups";fsi;
while (0<=size(R)-1){
a:=R[0];B:=B+min(count_eq(a,R),count_eq(a,L));
R:=remove(a,R);j:=j+1;
```

```
};
afficher(R0+", "+N+" N, "+(B-N)+" B");
}
return ("perdu " , L);
}:;
```

19.2.3 Le programme de la variante du Master-Mind

Il faut vérifier que la liste choisie n'est pas (6,6,6,6) avec la fonction : `nonconst(L,6)` ci dessous :

```
nonconst(L,a):={
local j,n:=size(L);
pour j de 0 jusque n-1 faire
si L[j]!=a alors return 1 fsi;
fpour;
return 0;
}
:;
mastervariante(nc):={
local L,R,R0,n,j,N,B,a;
repeter L:=[alea(7)$(j=1..4)]; jusqua nonconst(L,6);
n:=0;
while (n<nc) {
saisir(R);
R:=[R];n:=n+1;R0:=R;
N:=0;B:=0;j:=0;
pour j de 0 jusque 3 faire
a:=R[j];si a==L[j] alors N:=N+1; fsi;
fpour;
if (N==4) alors return "gagne en "+n+" coups";fsi;
while (0<=size(R)-1){
a:=R[0];B:=B+min(count_eq(a,R),count_eq(a,L));
R:=remove(a,R);j:=j+1;
};
afficher(R0+", "+N+" N, "+(B-N)+" B");
}
return ("perdu " , L);
}
:;
```

19.2.4 Le mastermind avec les couleurs

Le joueur devra maintenant taper les couleurs (ou juste sa première lettre) qu'il choisit (entourée de '"'). On tape pour transformer la liste des couleurs en la liste de leur code.

```
conv(L):={
  local d,j,R;
```

```
  d:=size(L);
  R:=makelist(0,1,d);
  pour j de 0 jusque d-1 faire
   if L[j][0]=="n" or L[j][0]=="N" alors R[j]:=0;elif
    L[j][0]=="r" or L[j][0]=="R" alors R[j]:=1;elif
     L[j][0]=="v" or L[j][0]=="V" alors R[j]:=2;elif
      L[j][0]=="j" or L[j][0]=="J" alors R[j]:=3;elif
       L[j][0]=="b" or L[j][0]=="B" alors R[j]:=4;elif
        L[j][0]=="m" or L[j][0]=="M" alors R[j]:=5;end;
  fpour;
  return R;
}:;
```

On tape pour transformer une liste de code couleur en des carrés de couleur, suivi de disques noir ou blanc qui donnent le résultat de l'essai.

```
L2C(l,n,N,B):={
local d,j,C;
C:=NULL;
d:=size(l);
pour j de 0 jusque d-1 faire
C:=C,affichage(carre(-5+j/4+(5-n/4)*i,-5+(j+1)/4+(5-n/4)*i),
                                              l[j]+rempli);
fpour;
pour j de 1 jusque N faire
C:=C,affichage(cercle(-5+(d)/4+j/4+(5-n/4+1/8)*i,0.1),rempli);
fpour;
pour j de 1 jusque B faire
C:=C,cercle(-5+(d+N+j)/4+(5-n/4+1/8)*i,0.08);
fpour;
return C;
}
:;
```

Dans le programme `mastermind(nc)`, le joueur devra taper une chaîne de 4 lettres sans les guillemets (") qui sont les initiales des noms des couleurs par exemple `rvnn` si vous péférez taper une liste de caractères par exemple `"r","v","n","n"` il faut décommenter `saisir(R); R:=conv([R]);` et commenter la ligne suivante. Le programme remplacera une lettre non valide par `"n"`. On tape si Le joueur peut faire au plus nc propositions :

```
mastermind(nc):={
local L,R,R0,n,j,N,B,a;
print("jeu du mastermind");
print("taper 4 couleurs parmi NRVJBM par ex RVNB");
gl_x=-5.3.2;
gl_y=-2..5.3;
ClrGraph();DispG();
L:=[alea(6)$(j=1..4)];
n:=0;
```

```
while (n<nc) {
//saisir(R); R:=conv([R]);
repeter saisir_chaine("4 couleurs parmi NRJVBM, ex. RRBJ",R); jusqua size(R)=
R:=conv(R);
n:=n+1;R0:=R;
N:=0;B:=0;j:=0;
pour j de 0 jusque 3 faire
a:=R[j];si a==L[j] alors N:=N+1; fsi;
fpour;
si (N==4) alors
L2C(R0,n,4,0),L2C(L,n+2,0,0);DispG();
return afficher("gagne en "+n+" coups");
fsi;
while (0<=size(R)-1){
a:=R[0];B:=B+min(count_eq(a,R),count_eq(a,L));
R:=remove(a,R);j:=j+1;
};
B:=B-N;
L2C(R0,n,N,B);DispG();
}
L2C(L,n+2,0,0);
return afficher(("perdu " , L));
}:;
```

A vous de programmer la variante avec les couleurs !

19.3 Les 4 dés du jeu de Win

On considère les 4 dés suivants :

- les faces du dé `A` ont comme points : `0,0,4,4,4,4`,
- les faces du dé `B` ont comme points : `1,1,1,5,5,5`,
- les faces du dé `C` ont comme points : `2,2,2,2,6,6`,
- les faces du dé `D` ont comme points : `3,3,3,3,3,3`,

La partie se compose de 12 lancers.
Pour une partie, chacun des joueurs choisit un dé. À chaque lancer, celui qui a le meilleur score marque 1 point. La partie se compose de 12 lancers.
On veut simuler ce jeu pour que l'on puisse jouer contre l'ordinateur. L'ordinateur tire au hasard un dé, vous donne son choix, puis vous choisisez un dé parmi les 3 dés qui restent. Puis vous jouez....
Quel dé faut-il choisir pour gagner contre l'ordinateur ?

On se reportera à la section 12.3 pour voir l'analyse du jeu.
On utilise les listes A,B,C,D pour representer chaque dé, et pour jouer, on tape :

```
jeuwin():={
local deo,dem,po,pm,scoro,j,A,B,C,D;
deo:=char(rand(4)+65);
print("j'ai choisi le de "+ deo);
```

```
A:=[0,0,4,4,4,4];
B:=[1,1,1,5,5,5];
C:=[2,2,2,2,6,6];
D:=[3,3,3,3,3,3];
deo:=expr(deo);
repeter saisir("votre choix",dem);
jusqua dem!=deo;
scoro:=0;
for (j:=0;j<12;j++){
po:=deo[rand(6)];
pm:=dem[rand(6)];
print(po,pm);
if (po>pm) scoro:=scoro+1;
}
return [scoro,12-scoro];
}
:;
```

Ma stratégie :
Si l'ordinateur choisit le dé A, je choisis le dé B,
Si l'ordinateur choisit le dé B, je choisis le dé C,
Si l'ordinateur choisit le dé C, je choisis le dé D,
Si l'ordinateur choisit le dé D, je choisis le dé A,
on tape `jeuwins()` qui joue selon cette stratégie.
Pour connaitre le score de n parties selon cette stratégie, on tape `jeuxwin(n)` :

```
jeuwins():={
local deo,dem,po,pm,scoro,j,A,B,C,D;
A:=[0,0,4,4,4,4];
B:=[1,1,1,5,5,5];
C:=[2,2,2,2,6,6];
D:=[3,3,3,3,3,3];
deo:=rand(4);
dem:=irem(deo+1,4);
scoro:=0;
deo:=expr(char(deo+65));
dem:=expr(char(dem+65));
for (j:=0;j<12;j++){
po:=deo[rand(6)];
pm:=dem[rand(6)];
if (po>pm) scoro:=scoro+1;
}
return [scoro,12-scoro];
}
:;
jeuxwin(n):={
local scoro,j;
scoro:=0;
pour j de 1 jusque n faire
```

```
scoro:=scoro+jeuwins()[0];
fpour;
return [scoro,12*n-scoro];
}:;
```

On tape :
`jeuxwin(200)`
On obtient :
`[836,1564]`
0n a $836/2400., 1564/2400. \sim 0.348333333333, 0.651666666667$ soit environ 1/3,2/3.

19.4 Un jeu de cartes

Un joueur reçoit 4 cartes d'un jeu de 32 cartes.
- si il a un carré il reçoit 5082 euros
- si il a un brelan (3 cartes de même hauteur) il reçoit 50 euros
- si il a deux paires il reçoit 45 euros
- si il a une paire il reçoit 5082 euros
- sinon il doit payer 10 euros

Ce jeu est-il équitable ?
Faire un programme qui permet de faire n parties de ce jeu.
Il y a `comb(32,4)=35960` mains possibles.

- IL y a `comb(8,1)=8` possibilités d'avoir un carré.
  La probabilité d'avoir un carré est donc : `8/35960=1/4495`
- IL y a `comb(8,1)*comb(4,3)*comb(28,1)=896` possibilités d'avoir un brelan.
  La probabilité d'avoir un brelan est donc : `896/35960=112/4495`
- IL y a `comb(8,1)*comb(4,2)*comb(7,1)*comb(4,2)/2=1008` possibilités d'avoir deux paires.
  La probabilité d'avoir deux paires est donc :`1008/35960=126/4495`
- IL y a `comb(8,1)*comb(4,2)*comb(28,1)*comb(24,1)/2=16128` (ou encore `comb(8,1)*comb(4,2)*(comb(28,2)-comb(7,1)*comb(4,2))=16128`) possibilités d'avoir une paire.
  La probabilité d'avoir une paire est donc :`16128/35960=2016/4495`
- On utilise la probabilité complémentaire de ne pas avoir ni carré, ni brelan, ni paires : `1-(1+112+126+2016)/4495=448/899`

Le jeu est équitable si l'espérance de gain est nulle.
On tape :
`(5082+112*50+126*45+3*2016)/4495-10*448/899`
On obtient `0`
Le jeu est donc équitable.
On tape le programme en representant une carte par un nombre de 0 à 31 0..7 pour les trèfles, 8..15 pour les carreaux, 16..23 pour les cœurs et 24..31 pour les piques.
On tire au hasard 4 nombres a, b, c, d différents parmi 0..31 (`hasard(4,0..31)`), on aura un carré si :
`irem(a,8)==irem(b,8)==irem(c,8)==irem(d,8)`

```
cartes(n):={
  local j,s,a,b,c,d,na,nb,nc,L;
  s:=0;
  pour j de 1 jusque n faire
    a,b,c,d:= irem(hasard(4,0,31),8);
    L:=[a,b,c,d];
    //print(L);
    na:=count(x->x==a,L);
    nb:=count(x->x==b,L);
    nc:=count(x->x==c,L);
    if na==4 alors s:=s+5082; elif
      na==3 or nb==3 alors s:=s+50; elif
        na==2 and nb==2 and nc==2 alors s:=s+45; elif
          na==1 and nb==1 and nc==1 alors s:=s-10;
           else
            s:=s+3;
          end;
        fpour;
        return s;}
      :;
```

On fait le graphe des gains pour p parties de n tirages.

```
gain(p,n):={
local G,k,s,c;
G:=NULL;
s:=0
pour k de 1 jusque p faire
c:=cartes(n);
//print(c);
G:=G,point(k,c);
s:=s+c;
fpour;
print (evalf(s/(p*n)));
return G;
}:;
```

Chapitre 20

Illusions d'optique

20.1 Illusion sur les longueurs

On tape dans un éditeur de programmes :

```
illusion11():={
local L;
L:=NULL;
L:=L,segment(-2,-2+4*i);
L:=L,segment(0,4*i);
L:=L,segment(-2,-2-sqrt(3)/2-i/2);
L:=L,segment(-2,-2+sqrt(3)/2-i/2);
L:=L,segment(-2+4*i,-2-sqrt(3)/2+4*i+i/2);
L:=L,segment(-2+4*i,-2+sqrt(3)/2+4*i+i/2);
L:=L,segment(0,-sqrt(3)/2+i/2);
L:=L,segment(0,sqrt(3)/2+i/2);
retourne L;
}:;
illusion12():={
local L;
L:=NULL;
L:=L,segment(4*i,-sqrt(3)/2+4*i-i/2);
L:=L,segment(4*i,sqrt(3)/2+4*i-i/2);
L:=L,segment(2,3+4*i);
L:=L,segment(2.5+i,4.5+i);
L:=L,segment(2.5+2*i,4.5+2*i);
L:=L,segment(5,4+4*i);
retourne L;
}:;
```

Puis on compile avec `OK` ou avec $F9$.
On tape dans un niveau de géométrie :
`illusion11();`
`illusion12();`
On obtient (les longueurs des segments parallèles ne semblent pas égales) :



20.2 Illusion sur les formes

On tape dans un éditeur de programmes :

```
illusion2():={
local j,L;
L:=NULL;
pour j de 1 jusque 13 faire
L:=L,cercle(0,2*j);
fpour;
L:=L,carre(-13*(1+i),13*(1-i));
retourne L;
}:;
```

Puis on compile avec `OK` ou avec $F9$.
On tape dans un niveau de géométrie :
`illusion2();`
On obtient un carré dont les côtés sembles incurvés :



20.3 Illusion de cordes torsadées

On tape dans un éditeur de programmes :

```
raies():={
local L,j,A,B,xa,ya;
L:=NULL;
xa:=-22;
pour j de 1 jusque 6 faire
A:=point(xa+i*22);
B:=point(xa+4+i*22);
L:=L,triangle(0,A,B,affichage=rempli);
L:=L,triangle(0,-A,-B,affichage=rempli);
xa:=xa+8;
fpour;
ya:=-18;
pour j de 1 jusque 5 faire
A:=point(-22+i*ya);
B:=point(-22+i*(ya+4));
L:=L,triangle(0,A,B,affichage=rempli);
L:=L,triangle(0,-A,-B,affichage=rempli);
ya:=ya+8;
fpour;
L:=L,affichage(carre(-2*(1+i),2*(1-i)),rempli);
retourne L;
}:;
bande():={
local L,j,xa,A,B,C,D;
```

```
L:=NULL;
xa:=-13;
A:=point(xa+i*11);
B:=point(xa+2+i*11);
C:=point((xa+6)*12/11.+i*12);
D:=point((xa+4)*12/11.+i*12);
L:=L,polygone(A,B,C,D,affichage=7+rempli);
L:=L,polygone(-A,-B,-C,-D,affichage=7+rempli);
xa:=xa+2;
pour j de 1 jusque 2 faire
A:=point(xa+i*11);
B:=point(xa+2+i*11);
C:=point((xa+6)*12/11.+i*12);
D:=point((xa+4)*12/11.+i*12);
L:=L,polygone(A,B,C,D,affichage=rempli);
L:=L,polygone(-A,-B,-C,-D,affichage=rempli);
xa:=xa+2;
A:=point(xa+i*11);
B:=point(xa+2+i*11);
C:=point((xa+6)*12/11.+i*12);
D:=point((xa+4)*12/11.+i*12);
L:=L,polygone(A,B,C,D,affichage=7+rempli);
L:=L,polygone(-A,-B,-C,-D,affichage=7+rempli);
xa:=xa+2;
fpour;
A:=point(xa+i*11);
B:=point(xa+2+i*11);
C:=point((xa+4)+i*11);
D:=point((xa+6)+i*11);
L:=L,polygone(A,B,i*11.25,C,D,i*11.75,affichage=rempli),
polygone(-A,-B,-i*11.25,-C,-D,-i*11.75,affichage=rempli);
xa:=3;
pour j de 1 jusque 2 faire
A:=point(xa+i*11);
B:=point(xa+2+i*11);
C:=point((xa-2)*12/11.+i*12);
D:=point((xa-4)*12/11.+i*12);
L:=L,polygone(A,B,C,D,affichage=7+rempli);
L:=L,polygone(-A,-B,-C,-D,affichage=7+rempli);
xa:=xa+2;
A:=point(xa+i*11);
B:=point(xa+2+i*11);
C:=point((xa-2)*12/11.+i*12);
D:=point((xa-4)*12/11.+i*12);
L:=L,polygone(A,B,C,D,affichage=rempli);
L:=L,polygone(-A,-B,-C,-D,affichage=rempli);
xa:=xa+2;
fpour;
```

```
A:=point(xa+i*11);
B:=point(xa+2+i*11);
C:=point((xa-2)*12/11.+i*12);
D:=point((xa-4)*12/11.+i*12);
L:=L,polygone(A,B,C,D,affichage=7+rempli);
L:=L,polygone(-A,-B,-C,-D,affichage=7+rempli);
retourne L;
}:;
bande1():={
local L,j,ya,A,B,C,D;
L:=NULL;
ya:=-9
pour j de 1 jusque 2 faire
A:=point(ya*i-11);
B:=point((ya+2)*i-11);
C:=point((ya-2)*i*12/11.-12);
D:=point((ya-4)*12/11.*i-12);
L:=L,polygone(A,B,C,D,affichage=rempli);
L:=L,polygone(-A,-B,-C,-D,affichage=rempli);
ya:=ya+2;
A:=point(ya*i-11);
B:=point((ya+2)*i-11);
C:=point((ya-2)*12/11.*i-12);
D:=point((ya-4)*12/11.*i-12);
L:=L,polygone(A,B,C,D,affichage=7+rempli);
L:=L,polygone(-A,-B,-C,-D,affichage=7+rempli);
ya:=ya+2;
fpour;
A:=point(ya*i-11);
B:=point((ya+2)*i-11);
C:=point((ya-2)*i*12/11.-12);
D:=point((ya-4)*12/11.*i+-12);
L:=L,polygone(A,B,C,D,affichage=rempli);
L:=L,polygone(-A,-B,-C,-D,affichage=rempli);
ya:=ya+2;
A:=point((ya+2)*i*12/11.-12);
B:=point((ya)*i*12/11.-12);
C:=point((ya-2)*i*12/11.-12);
D:=point((ya-4)*i*12/11.-12);
L:=L,polygone(A,B,-11.75,C,D,-11.25,affichage=7+rempli),
polygone(-A,-B,11.75,-C,-D,11.25,affichage=7+rempli);
ya:=1;
pour j de 1 jusque 2 faire
A:=point((ya-2)*i-11);
B:=point((ya)*i-11);
C:=point((ya+4)*12/11.*i-12);
D:=point((ya+2)*12/11.*i-12);
L:=L,polygone(A,B,C,D,affichage=rempli);
```

```
L:=L,polygone(-A,-B,-C,-D,affichage=rempli);
ya:=ya+2;
A:=point((ya-2)*i-11);
B:=point((ya)*i-11);
C:=point((ya+4)*12/11.*i-12);
D:=point((ya+2)*12/11.*i-12);
L:=L,polygone(A,B,C,D,affichage=7+rempli);
L:=L,polygone(-A,-B,-C,-D,affichage=7+rempli);
ya:=ya+2;
fpour;
A:=point((ya-2)*i-11);
B:=point(ya*i-11);
C:=point((ya+4)*12/11.*i-12);
D:=point((ya+2)*12/11.*i-12);
L:=L,polygone(A,B,C,D,affichage=rempli);
L:=L,polygone(-A,-B,-C,-D,affichage=rempli);
retourne L;
}:;
bande2():={
local L,j,ya,A,B,C,D;
L:=NULL;
ya:=-13;
A:=point((ya+1)*i+11.6);
B:=point((ya+2)*i+11);
C:=point((ya+6)*12/11.*i+12);
D:=point((ya+4)*12/11.*i+12);
L:=L,polygone(A,B,C,D,affichage=rempli);
L:=L,polygone(-A,-B,-C,-D,affichage=rempli);
ya:=ya+2;
pour j de 1 jusque 2 faire
A:=point(ya*i+11);
B:=point((ya+2)*i+11);
C:=point((ya+6)*i*12/11.+12);
D:=point((ya+4)*12/11.*i+12);
L:=L,polygone(A,B,C,D,affichage=7+rempli);
L:=L,polygone(-A,-B,-C,-D,affichage=7+rempli);
ya:=ya+2;
A:=point(ya*i+11);
B:=point((ya+2)*i+11);
C:=point((ya+6)*12/11.*i+12);
D:=point((ya+4)*12/11.*i+12);
L:=L,polygone(A,B,C,D,affichage=rempli);
L:=L,polygone(-A,-B,-C,-D,affichage=rempli);
ya:=ya+2;
fpour;
A:=point((ya)*i+11):;
B:=point((ya+2)*i+11):;
C:=point((ya+3)*i+11.75):;
```

```
D:=point((ya+3)*i+11.25):;
L:=L,polygone(B,A,C,D,affichage=7+rempli);
L:=L,polygone(-B,-A,-C,-D,affichage=7+rempli);
L:=L,polygone(D,11,B,affichage=rempli);
L:=L,polygone(-D,-11,-B,affichage=rempli);
retourne L,conj(L);
}:;
illusion31():={
retourne raies(),bande(),bande1();
}:;
illusion32():={
retourne raies(),bande(),bande2();
}:;
```

Puis on compile avec `OK` ou avec $F9$.
On tape dans un niveau de géométrie :
`illusion31();`
On obtient un carré dont les côtés sembles incurvés :



On tape dans un niveau de géométrie :
`illusion32();`
On obtient un carré dont les côtés sembles incurvés :



20.4 Illustration

```
lx(z0,r):={
  return(segment(z0+r*(-1-i),z0+r*(1+i)),
         segment(z0+r*(1-i),z0+r*(-1+i)));
}:;
lc(z0,r):={
  return (cercle(z0,r,pi/4,7*pi/4));
}:;
la(z0,r):={
  return(segment(z0+r*(-1-i),z0+r*i),
         segment(z0+r*(1-i),z0+r*i),
         segment(z0+r*-0.5,z0+r*0.5));
}:;
ls(z0,r):={
  return (segment(z0+r*(-1/2-i),z0-r*i),
          segment(z0+r*(1/2+i),z0+r*i),
          cercle(z0+r*i/2,r/2,pi/2,3*pi/2),
          cercle(z0-r*i/2,r/2,-pi/2,pi/2));
```

```
}:;
illusion32();
affichage(lx(0,1),4+line_width_5);
affichage(lc(3.4+0.6*i,0.5),4+line_width_5);
affichage(la(6+1.1*i,0.9),4+line_width_5);
affichage(ls(10+1.7*i,1.3),4+line_width_5);
affichage(symetrie(droite(x=0),lc(3.4+0.6*i,0.5)),4+line_width_5);
affichage(la(-6+1.1*i,0.9),4+line_width_5);
affichage(conj(ls(-10-1.7*i,1.3)),4+line_width_5);
```

On obtient :



Index

Table des matières